

ADOBE® ACTIONSCRIPT® 3.0

编程



© 2008 Adobe Systems Incorporated. 保留所有权利。

Adobe® Flash® 的 Adobe® ActionScript® 3.0 编程

如果本指南是随包括最终用户协议的软件分发的，那么，本指南及其所描述的软件将按照该许可协议提供，而且必须遵照该许可协议的条款来使用或复制。除非该许可协议允许，否则，未经 **Adobe Systems Incorporated** 的事先书面许可，不得以任何形式或任何方法（电子、机械、录制或其它方法）对本指南中的任何部分进行复制、存储到检索系统中或者进行传播。请注意，本指南中的内容受版权法保护，即使它不随包括最终用户许可协议的软件一起分发也是如此。

本指南中的任何内容仅供参考，可在不另行通知的情况下进行更改，并且不应视为 **Adobe Systems Incorporated** 对本指南内容构成任何承诺。对于本指南中信息性内容可能存在的任何错误或不准确的地方，**Adobe Systems Incorporated** 不承担任何责任或义务。

请注意，您可能希望包括在您的项目中的现有插图或图像可能受版权法的保护。如果在未经授权的情况下将这些材料合并到您的新作品中，则可能会侵害版权所有者的权利。请确保从版权所有者那里获得了任何必需的许可。

在范例模板中对公司名称的任何引用都仅用于演示目的，而绝不涉及任何实际的组织。

Adobe, the Adobe logo, Adobe AIR, ActionScript, Flash, Flash Lite, Flex, Flex Builder, MXML, and Pixel Bender are either registered trademarks or trademarks of Adobe Systems Incorporated in the United States and/or other countries.

ActiveX and Windows are either registered trademarks or trademarks of Microsoft Corporation in the United States and other countries. Macintosh is a trademark of Apple Inc., registered in the United States and other countries. Java is a trademark or registered trademark of Sun Microsystems, Inc. in the United States and other countries. All other trademarks are the property of their respective owners.

This product includes software developed by the Apache Software Foundation (<http://www.apache.org/>).

MPEG Layer-3 audio compression technology licensed by Fraunhofer IIS and Thomson Multimedia (<http://www.mp3licensing.com>)

Speech compression and decompression technology licensed from Nellymoser, Inc. (www.nellymoser.com).

Video compression and decompression is powered by On2 TrueMotion video technology. © 1992-2005 On2 Technologies, Inc. All Rights Reserved.
<http://www.on2.com>.

This product includes software developed by the OpenSymphony Group (<http://www.opensymphony.com>).

This product contains either BSAFE and/or TIPEM software by RSA Security, Inc.



Sorenson Spark™ video compression and decompression technology licensed from Sorenson Media, Inc.

Adobe Systems Incorporated, 345 Park Avenue, San Jose, California 95110, USA

Notice to U.S. government end users. The software and documentation are “Commercial Items,” as that term is defined at 48 C.F.R. §2.101, consisting of “Commercial Computer Software” and “Commercial Computer Software Documentation,” as such terms are used in 48 C.F.R. §12.212 or 48 C.F.R. §227.7202, as applicable. Consistent with 48 C.F.R. §12.212 or 48 C.F.R. §§227.7202-1 through 227.7202-4, as applicable, the Commercial Computer Software and Commercial Computer Software Documentation are being licensed to U.S. Government end users (a) only as Commercial items and (b) with only those rights as are granted to all other end users pursuant to the terms and conditions herein. Unpublished-rights reserved under the copyright laws of the United States. Adobe Systems Incorporated, 345 Park Avenue, San Jose, CA 95110-2704, USA. For U.S. Government End Users, Adobe agrees to comply with all applicable equal opportunity laws including, if appropriate, the provisions of Executive Order 11246, as amended, Section 402 of the Vietnam Era Veterans Readjustment Assistance Act of 1974 (38 USC 4212), and Section 503 of the Rehabilitation Act of 1973, as amended, and the regulations at 41 CFR Parts 60-1 through 60-60, 60-250 ,and 60-741. The affirmative action clause and regulations contained in the preceding sentence shall be incorporated by reference.

目录

第 1 章：关于本手册

使用本手册	1
访问 ActionScript 文档	1
ActionScript 学习资源	2

第 2 章：ActionScript 3.0 简介

关于 ActionScript	4
ActionScript 3.0 的优点	4
ActionScript 3.0 中的新增功能	4
与早期版本的兼容性	6

第 3 章：ActionScript 快速入门

编程基础	8
处理对象	10
常用编程元素	16
示例：动画公文包片段	18
使用 ActionScript 构建应用程序	20
创建自己的类	23
示例：创建基本应用程序	24
运行后续示例	29

第 4 章：ActionScript 语言及其语法

语言概述	31
对象和类	31
包和命名空间	32
变量	39
数据类型	42
语法	52
运算符	56
条件语句	61
循环	63
函数	65

第 5 章：ActionScript 中面向对象的编程

面向对象的编程基础知识	75
类	76
接口	87
继承	89
高级主题	96
示例：GeometricShapes	101

第 6 章：处理日期和时间

日期和时间基础知识	108
管理日历日期和时间	108
控制时间间隔	110
示例：简单的模拟时钟	113

第 7 章：处理字符串

字符串基础知识	116
创建字符串	117
length 属性	118
处理字符串中的字符	118
比较字符串	119
获取其它对象的字符串表示形式	119
连接字符串	119
在字符串中查找子字符串和模式	120
在大小写之间转换字符串	123
示例：ASCII 字符图	123

第 8 章：处理数组

数组基础知识	128
索引数组	129
关联数组	137
多维数组	140
克隆数组	142
高级主题	142
示例：PlayList	145

第 9 章：处理错误

错误处理基础知识	150
错误类型	152
ActionScript 3.0 中的错误处理	153
使用 Flash Player 和 AIR 的调试版	154
在应用程序中处理同步错误	155
创建自定义错误类	159
响应错误事件和状态	160
比较错误类	163
示例：CustomErrors 应用程序	166

第 10 章：使用正则表达式

正则表达式基础知识	171
正则表达式语法	172
对字符串使用正则表达式的方法	183
示例：Wiki 分析器	184

第 11 章：处理 XML

XML 基础知识	188
用于处理 XML 的 E4X 方法	191
XML 对象	192
XMLList 对象	194
初始化 XML 变量	195
组合和变换 XML 对象	196
遍历 XML 结构	197
使用 XML 命名空间	201
XML 类型转换	202
读取外部 XML 文档	203
示例：从 Internet 加载 RSS 数据	203

第 12 章：处理事件

事件处理基础知识	207
ActionScript 3.0 事件处理与早期版本事件处理的不同之处	209
事件流	210
事件对象	211
事件侦听器	214
示例：Alarm Clock	219

第 13 章：显示编程

显示编程的基础知识	224
核心显示类	228
显示列表方法的优点	229
处理显示对象	231
处理显示对象	241
对象动画	256
动态加载显示内容	257
示例：SpriteArranger	259

第 14 章：使用绘图 API

绘图 API 使用基础知识	265
了解 Graphics 类	266
绘制直线和曲线	266
使用内置方法绘制形状	268
创建渐变线条和填充	269
将 Math 类与绘制方法配合使用	273
使用绘图 API 进行动画处理	273
示例：Algorithmic Visual Generator	274
绘图 API 高级用法	276
绘制路径	277
定义缠绕规则	278

使用图形数据类	280
关于使用 drawTriangles()	282
第 15 章：处理几何结构	
几何学基础知识	283
使用 Point 对象	284
使用 Rectangle 对象	286
使用 Matrix 对象	289
例如：将矩阵转换用于显示对象	290
第 16 章：过滤显示对象	
过滤显示对象的基础知识	293
创建和应用滤镜	294
可用的显示滤镜	300
示例：Filter Workbench	313
第 17 章：使用 Pixel Bender 着色器	
Pixel Bender 着色器基础知识	319
加载或嵌入着色器	321
访问着色器元数据	322
指定着色器输入和参数值	322
使用着色器	327
第 18 章：处理影片剪辑	
影片剪辑基础知识	337
处理 MovieClip 对象	338
控制影片剪辑回放	338
使用 ActionScript 创建 MovieClip 对象	340
加载外部 SWF 文件	343
示例：RuntimeAssetsExplorer	344
第 19 章：使用补间动画	
补间动画基础知识	347
复制补间动画脚本	347
合并补间动画脚本	348
描述动画	349
添加滤镜	351
将补间动画与其显示对象关联	353
第 20 章：使用反向运动	
反向运动的基础知识	354
IK 骨架动画处理概述	355
获取有关 IK 骨架的信息	356
实例化 IKMover 并限制其移动	357

移动 IK 骨架	357
使用 IK 事件	358
第 21 章：处理文本	
处理文本的基本知识	359
使用 TextField 类	361
使用 Flash 文本引擎	380
第 22 章：处理位图	
处理位图的基本知识	405
Bitmap 和 BitmapData 类	407
处理像素	408
复制位图数据	410
使用杂点功能制作纹理	411
滚动位图	412
利用 mipmap 处理	413
示例：旋转的月球动画	413
第 23 章：在三维 (3D) 环境中工作	
3D 基础知识	423
了解 Flash Player 和 AIR 运行时的 3D 功能	424
创建和移动 3D 对象	425
将 3D 对象投影到 2D 视图上	427
示例：透视投影	428
执行复杂的 3D 转换	430
通过三角形获得 3D 效果	433
第 24 章：处理视频	
视频基础知识	440
了解视频格式	442
了解 Video 类	444
加载视频文件	444
控制视频播放	444
在全屏模式下播放视频	446
流式传输视频文件	449
了解提示点	449
编写元数据和提示点的回调方法	450
使用提示点和元数据	454
捕获摄像头输入	462
向服务器发送视频	467
FLV 文件的高级主题	467
示例：视频自动唱片点唱机	468

第 25 章：处理声音

声音处理基础知识	473
了解声音体系结构	475
加载外部声音文件	475
处理嵌入的声音	477
处理声音流文件	478
处理动态生成的音频	478
播放声音	480
加载和播放声音时的安全注意事项	483
控制音量和声相	483
处理声音元数据	484
访问原始声音数据	485
捕获声音输入	488
示例：Podcast Player	491

第 26 章：捕获用户输入

用户输入基础知识	498
捕获键盘输入	499
捕获鼠标输入	501
示例：WordSearch	504

第 27 章：网络与通信

网络和通信基础知识	508
处理外部数据	510
连接到其它 Flash Player 和 AIR 实例	515
套接字连接	519
存储本地数据	522
使用数据文件	524
示例：构建 Telnet 客户端	536
示例：上载和下载文件	539

第 28 章：客户端系统环境

客户端系统环境基础知识	544
使用 System 类	545
使用 Capabilities 类	546
使用 ApplicationDomain 类	547
使用 IME 类	549
示例：检测系统功能	553

第 29 章：复制和粘贴

复制和粘贴基础知识	557
读取和写入系统剪贴板	557
剪贴板数据格式	558

第 30 章：打印

打印基础知识	562
打印页面	563
Flash Player 和 AIR 任务和系统打印	564
设置大小、缩放和方向	566
示例：多页打印	567
示例：缩放、裁剪和拼接	569

第 31 章：使用外部 API

使用外部 API 的基础知识	571
外部 API 要求和优点	573
使用 ExternalInterface 类	573
示例：将外部 API 用于网页容器	576
示例：将外部 API 用于 ActiveX 容器	582

第 32 章：Flash Player 安全性

Flash Player 安全性概述	587
安全沙箱	588
权限控制	590
限制网络 API	595
全屏模式安全性	597
加载内容	598
跨脚本访问	600
作为数据访问加载的媒体	602
加载数据	604
从导入到安全域的 SWF 文件加载嵌入内容	606
处理旧内容	606
设置 LocalConnection 权限	606
控制外出 URL 访问	607
共享对象	608
摄像头、麦克风、剪贴板、鼠标和键盘访问	609
索引	610

第 1 章：关于本手册

本手册为在 Adobe® ActionScript® 3.0 中开发应用程序提供了基础。为了充分理解所介绍的理念和方法，您应已熟悉一般的编程概念，如数据类型、变量、循环和函数。您还应了解面向对象编程的基本概念，如类和继承。如果以前掌握了 ActionScript 1.0 或 ActionScript 2.0 知识，则会非常有帮助，但这并不是必需的。

使用本手册

本手册中的章节划分为以下逻辑组，以帮助您方便地查找 ActionScript 文档的相关部分：

章节	描述
第 2-5 章， ActionScript 编程概述	讨论核心 ActionScript 3.0 概念，包括语法规、语句和运算符以及面向对象的 ActionScript 编程。
第 6-11 章， ActionScript 3.0 核心数据类型和类	介绍 ActionScript 3.0 中的顶级数据类型。
第 12-32 章， Flash Player 和 Adobe AIR API	介绍在特定于 Adobe Flash Player 和 Adobe AIR 的包和类中实现的重要功能，其中包括事件处理、使用显示对象和显示列表、网络和通信、文件输入和输出、外部接口、应用程序安全模型等。

本手册还包含许多范例文件，这些文件演示重要类或常用类的应用程序编程概念。范例文件是以打包形式提供的，更加易于加载并能与 Adobe® Flash® CS4 Professional 一起使用，这些范例文件还可能包含包装文件。不过，核心范例代码是纯粹的 ActionScript 3.0，您可以在您喜爱的任何开发环境中使用。

可以使用多种方法来编写并编译 ActionScript 3.0，其中包括：

- 使用 Adobe Flex Builder 3 开发环境
- 使用任何文本编辑器和命令行编译器，如随 Flex Builder 3 提供的编辑器和编译器
- 使用 Adobe® Flash® CS4 Professional 创作工具

有关 ActionScript 开发环境的详细信息，请参阅 第 4 页的 “[ActionScript 3.0 简介](#)”

要理解本手册中的代码范例，您并不需要具备 ActionScript 集成开发环境（如 Flex Builder 或 Flash 创作工具）的相关使用经验。但是，您需要参考这些工具的文档，以了解如何使用它们来编写并编译 ActionScript 3.0 代码。有关详细信息，请参阅 第 1 页的 “[访问 ActionScript 文档](#)”。

访问 ActionScript 文档

因为本手册重点介绍 ActionScript 3.0（一种内容丰富且功能强大的面向对象编程语言），所以，它并未详细介绍特定工具或服务器体系结构内的应用程序开发过程或工作流程。因此，在设计、开发、测试和部署 ActionScript 3.0 应用程序时，除了《ActionScript 3.0 编程》外，您还需要查阅其它文档资料。

ActionScript 3.0 文档

本手册向您介绍 ActionScript 3.0 编程语言的概念，并提供实现详细信息以及阐释重要语言功能的范例。但是，本手册不是完整的语言参考。有关完整的语言参考，请参阅《ActionScript 3.0 语言和组件参考》，该手册介绍了语言中的每一个类、方法、属性和事件。《ActionScript 3.0 语言和组件参考》提供了关于核心语言、Flash 创作工具组件（在 fl 包中）以及 Flash Player 和 Adobe AIR API（在 Flash 包中）的详细参考信息。

Flash 文档

使用 Flash 创作工具时，可能需要参考以下手册：

书籍	描述
《使用 Flash》	描述如何在 Flash 创作工具中开发动态 Web 应用程序
《ActionScript 3.0 编程》	描述 ActionScript 3.0 语言以及核心 Flash Player 和 Adobe AIR API 的具体用法
《ActionScript 3.0 语言和组件参考》	提供 Flash 创作工具组件和 ActionScript 3.0 API 的语法、用法和代码示例
《使用 ActionScript 3.0 组件》	详细说明如何使用组件开发通过 Flash 创建的应用程序
《使用 Flash CS4 Professional 开发 Adobe AIR 应用程序》	介绍如何在 Flash 中使用 ActionScript 3.0 和 Adobe AIR API 开发和部署 Adobe AIR 应用程序。
《学习 Adobe Flash 中的 ActionScript 2.0》	提供 ActionScript 2.0 的语法概述并介绍在处理不同对象类型时如何使用 ActionScript 2.0
《ActionScript 2.0 语言参考》	提供 Flash 创作工具组件和 ActionScript 2.0 API 的语法、用法和代码示例
《使用 ActionScript 2.0 组件》	详细介绍如何使用 ActionScript 2.0 组件来开发通过 Flash 创建的应用程序
ActionScript 2.0 组件语言参考	描述第 2 版 Adobe 组件体系结构中提供的每个组件以及它的 API
扩展 Flash	描述 JavaScript API 中提供的对象、方法和属性
《Flash Lite 2.x 入门》	介绍如何使用 Adobe® Flash® Lite™ 2.x 开发应用程序，并提供可以与 Flash Lite 2.x 一起使用的 ActionScript 功能的语法、用法和代码示例
开发 Flash Lite 2.x 应用程序	介绍如何开发 Flash Lite 2.x 应用程序
《Flash Lite 2.x ActionScript 简介》	介绍如何使用 Flash Lite 2.x 来开发应用程序，并描述 Flash Lite 2.x 开发人员可以使用的所有 ActionScript 功能
Flash Lite 2.x ActionScript 语言参考	提供可以在 Flash Lite 2.x 中使用的 ActionScript 2.0 API 的语法、用法和代码示例
《Flash Lite 1.x 入门》	介绍 Flash Lite 1.x 并描述如何使用 Adobe® Device Central CS4 模拟器测试您的内容
开发 Flash Lite 1.x 应用程序	描述如何使用 Flash Lite 1.x 来开发用于移动设备的应用程序
学习 Flash Lite 1.x ActionScript	介绍如何在 Flash Lite 1.x 应用程序中使用 ActionScript，并描述可用于 Flash Lite 1.x 的所有 ActionScript 功能
Flash Lite 1.x ActionScript 语言参考	提供可用于 Flash Lite 1.x 的 ActionScript 元素的语法和用法

ActionScript 学习资源

除了这些手册中的内容以外，Adobe 还在 Adobe 开发人员中心和 Adobe 设计中心上提供定期更新的文章、设计思路和示例。

Adobe 开发人员中心

Adobe 开发人员中心提供有关 ActionScript 的最新信息、有关实际应用程序开发的文章以及新出现的重要问题的相关信息。开发人员中心的网址为 www.adobe.com/devnet/。

Adobe 设计中心

了解数字设计和动画图形方面的最新动态。您可以从中浏览主要艺术家的作品，了解新的设计趋势以及通过教程、重要工作流程和高级技巧来提高您的技能。请每月访问两次该网站以了解最新的教程和文章以及充满灵感的图片作品。设计中心的网址为 www.adobe.com/designcenter/。

第 2 章：ActionScript 3.0 简介

本章概述最新且最具创新性的 ActionScript 版本，即 Adobe® ActionScript® 3.0。

关于 ActionScript

ActionScript 是 Adobe® Flash® Player 和 Adobe® AIR™ 运行时环境的编程语言。它在 Flash、Flex 和 AIR 内容和应用程序中实现交互性、数据处理以及其它许多功能。

ActionScript 是由 Flash Player 和 AIR 中的 ActionScript 虚拟机 (AVM) 执行的。ActionScript 代码通常由编译器（如 Adobe® Flash® CS4 Professional 或 Adobe® Flex™ Builder™ 的内置编译器或 Adobe® Flex™ SDK 中提供的编辑器）编译为“字节代码格式”（一种由计算机编写并且计算机能够理解的编程语言）。字节码嵌入在 SWF 文件中，SWF 文件由 Flash Player 和 AIR 执行。

ActionScript 3.0 提供了可靠的编程模型，具备面向对象编程的基本知识的开发人员对此模型会感到似曾相识。ActionScript 3.0 相对于早期 ActionScript 版本改进的一些重要功能包括：

- 一个新增的 ActionScript 虚拟机，称为 AVM2，它使用全新的字节代码指令集，可使性能显著提高
- 一个更为先进的编译器代码库，可执行比早期编译器版本更深入的优化
- 一个扩展并改进的应用程序编程接口 (API)，拥有对对象的低级控制和真正意义上的面向对象的模型
- 一个基于 ECMA Script for XML (E4X) 规范 (ECMA-357 第 2 版) 的 XML API。E4X 是 ECMA Script 的一种语言扩展，它将 XML 添加为语言的本机数据类型。
- 一个基于文档对象模型 (DOM) 第 3 级事件规范的事件模型

ActionScript 3.0 的优点

ActionScript 3.0 的脚本编写功能超越了 ActionScript 的早期版本。它旨在方便创建拥有大型数据集和面向对象的可重用代码库的高度复杂应用程序。虽然 ActionScript 3.0 对于在 Adobe Flash Player 中运行的内容并不是必需的，但它使用新型的虚拟机 AVM2 实现了性能的改善。ActionScript 3.0 代码的执行速度可以比旧式 ActionScript 代码快 10 倍。

旧版本的 ActionScript 虚拟机 AVM1 执行 ActionScript 1.0 和 ActionScript 2.0 代码。为了向后兼容现有内容和旧内容，Flash Player 9 和 10 支持 AVM1。有关详细信息，请参阅第 6 页的“[与早期版本的兼容性](#)”。

ActionScript 3.0 中的新增功能

虽然 ActionScript 3.0 包含 ActionScript 编程人员所熟悉的许多类和功能，但 ActionScript 3.0 在架构和概念上是区别于早期的 ActionScript 版本的。ActionScript 3.0 中的改进部分包括新增的核心语言功能，以及能够更好地控制低级对象的改进 Flash Player API。

注：Adobe® AIR™ 应用程序还可以使用 Flash Player API。

核心语言功能

核心语言定义编程语言的基本构造块，例如语句、表达式、条件、循环和类型。 ActionScript 3.0 包含许多加速开发过程的新功能。

运行时异常

ActionScript 3.0 报告的错误情形比早期的 ActionScript 版本多。运行时异常用于常见的错误情形，可改善调试体验并使您能够开发可以可靠地处理错误的应用程序。运行时错误可提供带有源文件和行号信息注释的堆栈跟踪，以帮助您快速定位错误。

运行时类型

在 ActionScript 2.0 中，类型注释主要是为开发人员提供帮助；在运行时，所有值的类型都是动态指定的。在 ActionScript 3.0 中，类型信息在运行时保留，并可用于多种目的。Flash Player 和 Adobe AIR 执行运行时类型检查，从而改善了系统的类型安全性。类型信息还可用于以本机形式表示变量，从而提高了性能并减少了内存使用量。

密封类

ActionScript 3.0 引入了密封类的概念。密封类只能拥有在编译时定义的固定的一组属性和方法；不能添加其它属性和方法。这使得编译时的检查更为严格，从而导致程序更可靠。由于不要求每个对象实例都有一个内部哈希表，因此还提高了内存的使用率。还可以通过使用 `dynamic` 关键字来实现动态类。默认情况下，ActionScript 3.0 中的所有类都是密封的，但可以使用 `dynamic` 关键字将其声明为动态类。

闭包方法

ActionScript 3.0 使闭包方法可以自动记起它的原始对象实例。此功能对于事件处理非常有用。在 ActionScript 2.0 中，闭包方法无法记起它是从哪个对象实例提取的，所以在调用闭包方法时将导致意外的行为。`mx.utils.Delegate` 类是一种常用的解决方法，但已不再需要。

ECMAScript for XML (E4X)

ActionScript 3.0 实现了 ECMAScript for XML (E4X)，后者最近被标准化为 ECMA-357。E4X 提供一组用于操作 XML 的自然流畅的语言构造。与传统的 XML 分析 API 不同，使用 E4X 的 XML 就像该语言的本机数据类型一样执行。E4X 通过大大减少所需代码的数量来简化操作 XML 的应用程序的开发。有关 E4X 的 ActionScript 3.0 实现的详细信息，请参阅第 188 页的“[处理 XML](#)”。

若要查看 ECMA 的 E4X 规范，请访问 www.ecma-international.org。

正则表达式

ActionScript 3.0 包括对正则表达式的固有支持，因此您可以快速搜索并操作字符串。由于在 ECMAScript (ECMA-262) 第 3 版语言规范中对正则表达式进行了定义，因此 ActionScript 3.0 实现了对正则表达式的支持。

命名空间

命名空间与用于控制声明（`public`、`private`、`protected`）的可见性的传统访问说明符类似。它们的工作方式与名称由您指定的自定义访问说明符类似。命名空间使用统一资源标识符 (URI) 以避免冲突，而且在您使用 E4X 时还用于表示 XML 命名空间。

新基元类型

ActionScript 2.0 拥有单一数值类型 `Number`，它是一种双精度浮点数。ActionScript 3.0 包含 `int` 和 `uint` 类型。`int` 类型是一个带符号的 32 位整数，它使 ActionScript 代码可充分利用 CPU 的快速处理整数数学运算的能力。`int` 类型对使用整数的循环计数器和变量都非常有用。`uint` 类型是无符号的 32 位整数类型，可用于 RGB 颜色值、字节计数和其它方面。

Flash Player API 功能

ActionScript 3.0 中的 Flash Player API 包含许多用于在低级别控制对象的类。语言的体系结构的设计比早期版本更为直观。由于需要在这里详细介绍的新类实在太多，因此以下各节将着重介绍一些重要的更改。

注：Adobe® AIR™ 应用程序还可以使用 Flash Player API。

DOM3 事件模型

文档对象模型第 3 级事件模型 (DOM3) 提供了一种生成并处理事件消息的标准方法，以使应用程序中的对象可以进行交互和通信，同时保持自身的状态并响应更改。通过采用万维网联盟 DOM 第 3 级事件规范，该模型提供了一种比早期的 ActionScript 版本中所用的事件系统更清楚、更有效的机制。

事件和错误事件都位于 `flash.events` 包中。Flash 组件框架使用的事件模型与 Flash Player API 相同，因此事件系统在整个 Flash 平台中是统一的。

显示列表 API

用于访问 Flash Player 和 Adobe AIR 显示列表（包含应用程序中所有可视元素的树）的 API 由处理可视基元的类组成。

新增的 `Sprite` 类是一个轻型构造块，它类似于 `MovieClip` 类，但更适合作为 UI 组件的基类。新增的 `Shape` 类表示原始的矢量形状。可以使用 `new` 运算符很自然地实例化这些类，并可以随时动态地重新指定其父类。

现在，深度管理是自动执行的并且已内置于 Flash Player 和 Adobe AIR 中，因此不需要指定深度编号。提供了用于指定和管理对象的 `z` 顺序的新方法。

处理动态数据和内容

ActionScript 3.0 包含用于加载和处理应用程序中的资源和数据的机制，这些机制在 API 中是直观的并且是一致的。新增的 `Loader` 类提供了一种加载 SWF 文件和图像资源的单一机制，并提供了一种访问已加载内容的详细信息的方法。`URLLoader` 类提供了一种单独的机制，用于在数据驱动的应用程序中加载文本和二进制数据。`Socket` 类提供了一种以任意格式从 / 向服务器套接字中读取 / 写入二进制数据的方式。

低级数据访问

各种 API 提供了对数据的低级访问，而这种访问以前在 ActionScript 中是不可能的。对于正在下载的数据而言，可使用 `URLStream` 类（由 `URLLoader` 实现）在下载数据的同时访问原始二进制数据。使用 `ByteArray` 类可优化二进制数据的读取、写入以及处理。使用新增的 Sound API，可以通过 `SoundChannel` 类和 `SoundMixer` 类对声音进行精细控制。新增的处理安全性的 API 可提供有关 SWF 文件或加载内容的安全权限的信息，从而使您能够更好地处理安全错误。

处理文本

ActionScript 3.0 包含一个用于所有与文本相关的 API 的 `flash.text` 包。`TextLineMetrics` 类为文本字段中的一行文本提供精确度量；该类取代了 ActionScript 2.0 中的 `TextFormat.getTextExtent()` 方法。`TextField` 类包含许多有趣的新低级方法，这些方法可以提供有关文本字段中的一行文本或单个字符的特定信息。这些方法包括 `getCharBoundaries()`（返回一个表示字符边框的矩形）、`getCharIndexAtPoint()`（返回指定点处字符的索引）以及 `getFirstCharInParagraph()`（返回段落中第一个字符的索引）。行级方法包括 `getLineLength()`（返回指定文本行中的字符数）和 `getLineText()`（返回指定行的文本）。新增的 `Font` 类提供了一种管理 SWF 文件中的嵌入字体的方法。

与早期版本的兼容性

和以往一样，Flash Player 提供针对以前发布的内容的完全向后兼容性。可在早期版本的 Flash Player 中运行的任何内容都可在 Flash Player 9 及更高版本中运行。然而，在 Flash Player 9 中引入 ActionScript 3.0 后，的确对在 Flash Player 9 或更高版本中运行的旧内容和新内容之间的互操作性提出了一些挑战。兼容性问题包括以下几个方面：

- 单个 SWF 文件无法将 ActionScript 1.0 或 2.0 代码和 ActionScript 3.0 代码组合在一起。

- ActionScript 3.0 代码可以加载以 ActionScript 1.0 或 2.0 编写的 SWF 文件，但无法访问 SWF 文件的变量和函数。
- 以 ActionScript 1.0 或 2.0 编写的 SWF 文件无法加载以 ActionScript 3.0 编写的 SWF 文件。这意味着，在 Flash 8 或 Flex Builder 1.5 或更早版本中创作的 SWF 文件无法加载 ActionScript 3.0 SWF 文件。

此规则的唯一例外情况是，只要 ActionScript 2.0 SWF 文件以前没有向它的任何级别加载任何内容，ActionScript 2.0 SWF 文件就可以用 ActionScript 3.0 SWF 文件来替换它自身。ActionScript 2.0 SWF 文件可通过调用 `loadMovieNum()` 并将值 0 传递给 `level` 参数来实现此目的。

- 通常，以 ActionScript 1.0 或 2.0 编写的 SWF 文件必须进行迁移才能与以 ActionScript 3.0 编写的 SWF 文件一起使用。例如，假定使用 ActionScript 2.0 创建了一个媒体播放器。该媒体播放器加载同样是使用 ActionScript 2.0 创建的各种内容。您不能在该媒体播放器中加载以 ActionScript 3.0 创建的新内容。您必须将视频播放器迁移到 ActionScript 3.0。

但是，如果您在 ActionScript 3.0 中创建一个媒体播放器，则该媒体播放器可以执行 ActionScript 2.0 内容的简单加载。

下面这些表总结了早期 Flash Player 版本在加载新内容和执行代码方面的局限性，以及在不同的 ActionScript 版本中编写的 SWF 文件之间跨脚本编写的局限性。

支持的功能	Flash Player 7	Flash Player 8	Flash Player 9 和 10
可以加载针对以下版本发布的 SWF	7 和更早版本	8 和更早版本	9 (或 10) 及更低版本
包含此 AVM	AVM1	AVM1	AVM1 和 AVM2
运行在以下 ActionScript 版本中编写的 SWF	1.0 和 2.0	1.0 和 2.0	1.0、2.0 和 3.0

在下表中，“支持的功能”指在 Flash Player 9 或更高版本中运行的内容。运行在 Flash Player 8 或更早版本中的内容只能在 ActionScript 1.0 和 2.0 中加载、显示、执行以及跨脚本编写。

支持的功能	以 ActionScript 1.0 和 2.0 创建的内容	以 ActionScript 3.0 创建的内容
可以加载在以下版本中创建的内容并在其中执行代码	仅 ActionScript 1.0 和 2.0	ActionScript 1.0、2.0 和 ActionScript 3.0
可以对在以下版本中创建的内容进行跨脚本编写	仅 ActionScript 1.0 和 2.0 (通过本地连接的 ActionScript 3.0)	ActionScript 1.0 和 2.0 (通过本地连接)。 ActionScript 3.0

第 3 章：ActionScript 快速入门

本章旨在让您着手进行 ActionScript 编程，同时为您提供了解本手册其余部分的概念和示例所需的背景。本章首先讨论基本的编程概念，这些概念在关于如何在 ActionScript 中应用它们的上下文中介绍。本章还介绍了关于如何组织和构建 ActionScript 应用程序的要点。

编程基础

因为 ActionScript 是一种编程语言，所以，如果您首先了解几个通用的计算机编程概念，则会对您学习 ActionScript 很有帮助。

计算机程序的用途

首先，对计算机程序的概念及其用途有一个概念性的认识是非常有用的。计算机程序主要包括两个方面：

- 程序是计算机执行的一系列指令或步骤。
- 每一步最终都涉及到对某一段信息或数据的处理。

通常认为，计算机程序只是您提供给计算机并让它逐步执行的指令列表。每个单独的指令都称为语句。正如您将在本手册中看到的那样，在 ActionScript 中编写的每个语句的末尾都有一个分号。

实质上，程序中指令所做的全部事情是操作存储在计算机内存中的一些数据位。举一个简单的例子，您可能指示计算机将两个数字相加并将结果存储在计算机的内存中。举一个较复杂的例子：假设在屏幕上绘制了一个矩形，您希望编写一个程序将它移动到屏幕上的其它位置。计算机跟踪该矩形的某些信息 — 该矩形所在位置的 x 和 y 光标、其宽度和高度以及颜色等等。这些信息位中的每一位都存储在计算机内存中的某个位置。为了将矩形移动到其它位置，程序将采取类似于“将 x 坐标更改为 200；将 y 坐标更改为 150”的步骤（也就是说，为 x 和 y 坐标指定新值）。当然，计算机的确会对这些数据进行某些处理，以便切实地将这些数字转变为显示在计算机屏幕上的图像；但考虑到我们所感兴趣的详细程度，我们只要知道“在屏幕上移动矩形”这一过程确实只涉及更改计算机内存中的数据位就足够了。

变量和常量

由于编程主要涉及更改计算机内存中的信息，因此在程序中需要一种方法来表示单条信息。变量是一个名称，表示计算机内存中的值。在编写语句来操作值时，编写变量名来代替值；只要计算机看到程序中的变量名，就会查看自己的内存并使用在内存中找到的值。例如，如果两个名为 value1 和 value2 的变量分别包含一个数字，则可以编写如下语句将这两个数字相加：

```
value1 + value2
```

在实际执行这些步骤时，计算机会查看每个变量中的值，并将这些值相加。

在 ActionScript 3.0 中，一个变量实际上包含三个不同部分：

- 变量的名称
- 可以存储在变量中的数据的类型
- 存储在计算机内存中的实际值

刚才我们讨论了计算机如何将名称用作值的占位符。数据类型也非常重要。在 ActionScript 中创建变量时，应指定该变量要保存的数据的特定类型；此后，程序的指令只能在该变量中存储该类型的数据，您可以使用与该变量的数据类型关联的特定特性来操作值。在 ActionScript 中，若要创建一个变量（称为声明变量），应使用 var 语句：

```
var value1:Number;
```

在本例中，我们指示计算机创建一个名为 `value1` 的变量，该变量仅保存 Number 数据（“Number”是在 ActionScript 中定义的一种特定数据类型）。您还可以立即在变量中存储一个值：

```
var value2:Number = 17;
```

在 Adobe Flash CS4 Professional 中，还有另外一种变量声明方法。在将一个影片剪辑元件、按钮元件或文本字段放置在舞台上时，可以在“属性”检查器中为它指定一个实例名称。在后台，Flash 将创建一个与该实例名称同名的变量，您可以在 ActionScript 代码中使用该变量来引用该舞台项目。例如，如果您将一个影片剪辑元件放在舞台上并为它指定了实例名称 `rocketShip`，那么，只要您在 ActionScript 代码中使用变量 `rocketShip`，实际上就是在处理该影片剪辑。

常量也是一个名称，表示计算机内存中具有指定数据类型的值，就这一点而言，常量与变量极为相似。不同之处在于，在 ActionScript 应用程序运行期间只能为常量赋值一次。一旦为某个常量赋值之后，该常量的值在整个应用程序运行期间都保持不变。常量声明语法与变量声明语法相同，只不过是使用 `const` 关键字而不使用 `var` 关键字：

```
const SALES_TAX_RATE:Number = 0.07;
```

常量可用于定义在项目内多个位置使用的值，并且此值在正常情况下不会更改。使用常量而不使用字面值能让代码更加便于理解。例如，价格乘以 `SALES_TAX_RATE` 的代码行相对于价格乘以 `0.07` 的代码行更加易于理解。此外，如果确实需要更改通过常量定义的值，如果在整个项目中使用常量表示该值，则只需在一个位置（常量声明）更改该值，不需像使用硬编码字面值那样在不同位置更改该值。

数据类型

在 ActionScript 中，您可以将很多数据类型用作所创建的变量的数据类型。其中的某些数据类型可以看作是“简单”或“基本”数据类型：

- **String:** 一个文本值，例如，一个名称或书中某一段的文字
- **Numeric:** 对于 numeric 型数据，ActionScript 3.0 包含三种特定的数据类型：
 - **Number:** 任何数值，包括有小数部分或没有小数部分的值
 - **Int:** 一个整数（不带小数部分的整数）
 - **Uint:** 一个“无符号”整数，即不能为负数的整数
- **Boolean:** 一个 `true` 或 `false` 值，例如开关是否开启或两个值是否相等

简单数据类型表示单条信息：例如，单个数字或单个文本序列。然而，ActionScript 中定义的大部分数据类型都可以被描述为复杂数据类型，因为它们表示组合在一起的一组值。例如，数据类型为 `Date` 的变量表示一个值 — 某个时间时刻。然而，该日期值实际上表示为几个值：年、月、日、时、分、秒等等，它们都是单独的数字。所以，虽然我们认为日期是单个值（可以通过创建一个 `Date` 变量将日期作为单个值来对待），而在计算机内部却认为日期是组合在一起、共同定义单个日期的一组值。

大部分内置数据类型以及程序员定义的数据类型都是复杂数据类型。您可能认识下面的一些复杂数据类型：

- **MovieClip:** 影片剪辑元件
- **TextField:** 动态文本字段或输入文本字段
- **SimpleButton:** 按钮元件
- **Date:** 有关时间中的某个片刻的信息（日期和时间）

经常用作数据类型的同义词的两个词是类和对象。类仅仅是数据类型的定义 — 好比用于该数据类型的所有对象的模板，例如“所有 Example 数据类型的变量都拥有这些特性：A、B 和 C”。相反，对象仅仅是类的一个实际的实例；可将一个数据类型为 `MovieClip` 的变量描述为一个 `MovieClip` 对象。下面几条陈述虽然表达的方式不同，但意思是相同的：

- 变量 `myVariable` 的数据类型是 `Number`。
- 变量 `myVariable` 是一个 `Number` 实例。
- 变量 `myVariable` 是一个 `Number` 对象。
- 变量 `myVariable` 是 `Number` 类的一个实例。

处理对象

ActionScript 是一种面向对象的编程语言。面向对象的编程仅仅是一种编程方法，与使用对象来组织程序中的代码的方法实际上没有什么差别。

先前我们将计算机程序定义为计算机执行的一系列步骤或指令。那么从概念上讲，我们可能认为计算机程序只是一个很长的指令列表。然而，在面向对象的编程中，程序指令划分为不同的对象 — 代码分组为功能块，因此相关类型的功能或相关的信息会组合到一个容器中。

事实上，如果您已经在 Flash 中处理过元件，那么您应已习惯于使用对象了。假定您定义了一个影片剪辑元件 — 比如说，绘制一个矩形 — 并且您将该元件的副本放在了舞台上。从严格意义上来说，该影片剪辑元件也是 ActionScript 中的一个对象；即 MovieClip 类的一个实例。

您可以修改该影片剪辑的不同特征。例如，当选中该影片剪辑时，您可以在属性检查器中更改许多值，如该影片剪辑的 x 坐标、剪辑的宽度，也可以实施各种颜色调整（如更改剪辑的 alpha 值，即透明度），或对之应用投影滤镜。还可以使用其它 Flash 工具进行更多更改，例如，使用“任意变形”工具旋转该矩形。在 Flash 创作环境中修改一个影片剪辑元件时所做的更改，同样可在 ActionScript 中通过更改组合在一起、构成称为 MovieClip 对象的单个包的各数据片断来实现。

在 ActionScript 面向对象的编程中，任何类都可以包含三种类型的特性：

- 属性
- 方法
- 事件

这些元素共同用于管理程序使用的数据块，并用于确定执行哪些动作以及动作的执行顺序。

属性

属性表示某个对象中绑定在一起的若干数据块中的一个。Song 对象可能具有名为 artist 和 title 的属性；MovieClip 类具有 rotation、x、width 和 alpha 等属性。您可以像使用单个变量那样使用属性；事实上，可以将属性简单地视为包含在对象中的“子”变量。

以下是一些使用属性的 ActionScript 代码的示例。以下代码行将名为 square 的 MovieClip 移动到 100 个像素的 x 坐标处：

```
square.x = 100;
```

此代码使用 rotation 属性旋转 square MovieClip，以便与 triangle MovieClip 的旋转相匹配：

```
square.rotation = triangle.rotation;
```

此代码更改 square MovieClip 的水平缩放比例，以使其宽度为原始宽度的 1.5 倍：

```
square.scaleX = 1.5;
```

请注意上面几个示例的通用结构：将变量（square 和 triangle）用作对象的名称，后跟一个句点（.）和属性名（x、rotation 和 scaleX）。句点称为点运算符，用于指示您要访问对象的某个子元素。整个结构“变量名 - 点 - 属性名”的使用类似于单个变量，作为计算机内存中的单个值的名称。

方法

方法是指可以由对象执行的操作。例如，如果在 Flash 中使用时间轴上的几个关键帧和动画制作了一个影片剪辑元件，则可以播放或停止该影片剪辑，或者指示它将播放头移到特定的帧。

下面的代码指示名为 shortFilm 的 MovieClip 开始播放：

```
shortFilm.play();
```

下面的代码使名为 shortFilm 的 MovieClip 停止播放（播放头停在原地，就像暂停播放视频一样）：

```
shortFilm.stop();
```

下面的代码使名为 shortFilm 的 MovieClip 将其播放头移到第 1 帧，然后停止播放（就像后退视频一样）：

```
shortFilm.gotoAndStop(1);
```

正如您所看到的一样，您可以通过依次写下对象名（变量）、句点、方法名和小括号来访问方法，这与属性类似。小括号是指示要调用方法（即指示对象执行该操作）的方式。有时，为了传递执行动作所需的额外信息，将值（或变量）放入小括号中。这些值称为方法参数。例如，`gotoAndStop()` 方法需要知道应转到哪一帧，所以要求小括号中有一个参数。有些方法（如 `play()` 和 `stop()`）自身的意义已非常明确，因此不需要额外信息。但书写时仍然带有小括号。

与属性（和变量）不同的是，方法不能用作值占位符。然而，一些方法可以执行计算并返回可以像变量一样使用的结果。例如，`Number` 类的 `toString()` 方法将数值转换为文本表示形式：

```
var numericData:Number = 9;
var textData:String = numericData.toString();
```

例如，如果希望在屏幕上的文本字段中显示 `Number` 变量的值，应使用 `toString()` 方法。`TextField` 类的 `text` 属性（表示在屏幕上显示的实际文本内容）被定义为 `String`，所以只能包含文本值。下面的一行代码将变量 `numericData` 中的数值转换为文本，然后使这些文本显示在屏幕上名为 `calculatorDisplay` 的 `TextField` 对象中：

```
calculatorDisplay.text = numericData.toString();
```

事件

我们已经介绍了计算机程序就是计算机分步执行的一系列指令。一些简单的计算机程序仅包括计算机执行的几个步骤以及程序的结束点。然而，`ActionScript` 程序可以保持运行、等待用户输入或等待其它事件发生。事件是确定计算机执行哪些指令以及何时执行的机制。

本质上，事件就是所发生的、`ActionScript` 能够识别并可响应的事情。许多事件与用户交互有关 — 如用户单击按钮或按键盘上的键，但也有其它类型的事件。例如，如果使用 `ActionScript` 加载外部图像，有一个事件可让您知道图像何时加载完毕。本质上，当 `ActionScript` 程序正在运行时，`Adobe Flash Player` 和 `Adobe AIR` 只是坐等某些事情的发生，当这些事情发生时，它们将运行您为这些事件指定的特定 `ActionScript` 代码。

基本事件处理

用于指定为响应特定事件而应执行的特定操作的技术称为事件处理。在编写执行事件处理的 `ActionScript` 代码时，您需要识别三个重要元素：

- 事件源：发生该事件的是哪个对象？例如，哪个按钮会被单击，或哪个 `Loader` 对象正在加载图像？事件源也称为事件目标，因为 `Flash Player` 或 `AIR` 将此对象作为事件的目标（即事件实际发生的位置）。
- 事件：将要发生什么事情，以及您希望响应什么事情？识别事件是非常重要的，因为许多对象都会触发多个事件。
- 响应：当事件发生时，您希望执行哪些步骤？

无论何时编写处理事件的 `ActionScript` 代码，都会包括这三个元素，并且代码将遵循以下基本结构（以粗体显示的元素是您将针对具体情况填写的占位符）：

```
function eventResponse(eventObject:EventType):void
{
    // Actions performed in response to the event go here.
}

eventSource.addEventListener(EventType.EVENT_NAME, eventResponse);
```

此代码完成两项任务。首先，定义一个函数，这是指定为响应事件而要执行的动作的方法。接下来，调用源对象的 `addEventListener()` 方法，实际上就是为指定事件“订阅”该函数，以便当该事件发生时，执行该函数的操作。我们将更为详细地讨论其中每个部分。

函数提供一种将若干个操作组合在一起、用类似于快捷名称的单个名称来执行这些操作的方法。函数与方法完全相同，只是不必与特定类关联（事实上，可以将方法定义为与特定类关联的函数）。在创建用于事件处理的函数时，必须选择函数名称（本示例中名为 eventResponse）。还必须指定一个参数（本示例中名为 eventObject）。指定函数参数类似于声明变量，所以还必须指明参数的数据类型。（在本示例中，该参数的数据类型为 EventType。）

您要侦听的每种事件类型都有一个与其相关联的 ActionScript 类。为函数参数指定的数据类型始终是与您要响应的特定事件关联的类。例如，click 事件（在用户使用鼠标单击某个项目时触发）与 MouseEvent 类相关联。若要为 click 事件编写侦听器函数，可使用数据类型为 MouseEvent 的参数定义侦听器函数。最后，在左大括号与右大括号之间 ({ ... }) 编写希望计算机在事件发生时执行的指令。

一旦编写了事件处理函数，就需要告知事件源对象（发生事件的对象，如按钮）您希望在该事件发生时调用函数。可通过调用该对象的 addEventListener() 方法来实现此目的（所有具有事件的对象都同时具有 addEventListener() 方法）。

addEventListener() 方法有两个参数：

- 第一个参数是您希望响应的特定事件的名称。同样，每个事件都与一个特定类关联，而该类会为每个事件预定义一个特殊值（类似于事件自己的唯一名称，应将其用于第一个参数）。
- 第二个参数是事件响应函数的名称。请注意，如果将函数名称作为参数进行传递，则在写入函数名称时不使用括号。

了解事件处理过程

下面分步描述了创建事件侦听器时执行的过程。在本例中，您将创建一个侦听器函数，在单击名为 myButton 的对象时调用该函数。

程序员实际编写的代码如下所示：

```
function eventResponse(event:MouseEvent):void
{
    // Actions performed in response to the event go here.
}


myButton.addEventListener(MouseEvent.CLICK, eventResponse);
```

下面是此代码在 Flash Player 中运行时的实际工作方式。（该行为对于 Adobe AIR 是完全相同的）：


- 1 加载 SWF 文件时，Flash Player 会注意到有一个名为 eventResponse() 的函数。




- 2 Flash Player 随后运行该代码（具体地说，是指不在函数中的代码行）。在本例中，只有一行代码：针对事件源对象（名为 myButton）调用 addEventListener() 方法，并将 eventResponse 函数作为参数进行传递。



- a 在内部，myButton 包含正在侦听其每个事件的函数的列表，因此，当调用其 addEventListener() 方法时，myButton 将 eventResponse() 函数存储在其事件侦听器列表中。




- 3 在某一时刻，用户单击 myButton 对象以触发其 click 事件（在代码中将其标识为 MouseEvent.CLICK）。




此时发生了以下事件：


- a Flash Player 创建一个对象，它是与所述事件（本示例中为 MouseEvent）关联的类的实例。对于很多事件，这是 Event 类的实例；对于鼠标事件，这是 MouseEvent 实例；对于其它事件，这是与该事件关联的类的实例。创建的此对象称为事件对象，包含有关所发生的事件的特定信息：事件类型、发生位置以及其它特定于事件的信息（如果适用）。



- b Flash Player 随后查看 myButton 存储的事件侦听器的列表。它逐个查看这些函数，以调用每个函数并将事件对象作为参数传递给该函数。由于 eventResponse() 函数是 myButton 的侦听器之一，因此，Flash Player 将在此过程中调用 eventResponse() 函数。



- c 当调用 eventResponse() 函数时，将运行该函数中的代码，因此，将执行您指定的动作。



事件处理示例

以下几个更为具体的事件示例，让您对在编写事件处理代码时常用的一些事件元素以及可能的变化形式有一个了解：

- 单击按钮开始播放当前的影片剪辑。在下面的示例中，playButton 是按钮的实例名称，而 this 是表示“当前对象”的特殊名称：

```
this.stop();

function playMovie(event:MouseEvent):void
{
    this.play();
}

playButton.addEventListener(MouseEvent.CLICK, playMovie);
```

- 检测文本字段中的键入操作。在下面的示例中，entryText 是一个输入文本字段，而 outputText 是一个动态文本字段：

```
function updateOutput(event:TextEvent):void
{
    var pressedKey:String = event.text;
    outputText.text = "You typed: " + pressedKey;
}

entryText.addEventListener(TextEvent.TEXT_INPUT, updateOutput);
```

- 单击按钮导航到一个 URL。在本例中，linkButton 是该按钮的实例名称：

```
function gotoAdobeSite(event:MouseEvent):void
{
    var adobeURL:URLRequest = new URLRequest("http://www.adobe.com/");
    navigateToURL(adobeURL);
}

linkButton.addEventListener(MouseEvent.CLICK, gotoAdobeSite);
```

创建对象实例

当然，在 ActionScript 中使用对象之前，该对象首先必须存在。创建对象的步骤之一是声明变量；然而，声明变量仅仅是在计算机的内存中创建一个空位置。在尝试使用或操作变量之前，您必须为变量指定实际值（即创建一个对象并将之存储在变量中）。创建对象的过程称为实例化 对象，也就是创建特定类的实例。

有一种创建对象实例的简单方法完全不涉及 ActionScript。在 Flash 中，当将一个影片剪辑元件、按钮元件或文本字段放置在舞台上，并在“属性”检查器中为它指定实例名时，Flash 会自动声明一个拥有该实例名的变量、创建一个对象实例并将该对象存储在该变量中。同样，在 Adobe Flex Builder 中，当您在 MXML 中创建一个组件（通过用 MXML 标签进行编码或通过将组件放置在处于设计模式下的编辑器中）并为该组件分配一个 ID（在 MXML 标记中或在 Flex 属性视图中）时，该 ID 会成为一个 ActionScript 变量的名称，并且会创建该组件的一个实例并将之存储在该变量中。

然而，您不会总是希望直观地创建对象。还可以通过几种方法来仅使用 ActionScript 创建对象实例。首先，借助几个 ActionScript 数据类型，可以使用文本表达式（直接写入 ActionScript 代码的值）创建一个实例。下面给出了一些示例：

- 文本数字值（直接输入数字）：

```
var someNumber:Number = 17.239;
var someNegativeInteger:int = -53;
var someUint:uint = 22;
```

- 文本字符串值（用双引号将文本引起来）：

```
var firstName:String = "George";
var soliloquy:String = "To be or not to be, that is the question...";
```

- 文本布尔值（使用字面值 true 或 false）：

```
var niceWeather:Boolean = true;
var playingOutside:Boolean = false;
```

- 文本数组值（在方括号中包含以逗号分隔的值列表）：

```
var seasons:Array = ["spring", "summer", "autumn", "winter"];
```

- 文本 XML 值（直接输入 XML）：

```
var employee:XML = <employee>
    <firstName>Harold</firstName>
    <lastName>Webster</lastName>
</employee>;
```

ActionScript 还为 Array、RegExp、Object 和 Function 数据类型定义了文本表达式。有关这些类的详细信息，请参阅第 128 页的“[处理数组](#)”、第 171 页的“[使用正则表达式](#)”和第 48 页的“[Object 数据类型](#)”。

对于其它任何数据类型而言，要创建一个对象实例，应将 new 运算符与类名一起使用，如下所示：

```
var raceCar:MovieClip = new MovieClip();
var birthday>Date = new Date(2006, 7, 9);
```

使用 new 运算符创建对象通常称为“调用类的构造函数”。构造函数是在创建类的实例的过程中调用的一种特殊方法。请注意，当以此方法创建实例时，请在类名后加上小括号，有时还可以指定参数值 — 调用方法的同时还需执行这两项操作。

请注意，甚至对于可使用文本表达式创建实例的数据类型，也可以使用 new 运算符来创建对象实例。例如，下面的两行代码执行的是相同的操作：

```
var someNumber:Number = 6.33;
var someNumber:Number = new Number(6.33);
```

熟悉使用 new ClassName() 创建对象的方法是非常重要的。如果需要创建无可视化表示形式（因而无法通过将项目放置在 Flash 舞台上创建，也无法在 Flex Builder 的 MXML 编辑器的设计模式下创建）的 ActionScript 数据类型的一个实例，则只能通过使用 new 运算符在 ActionScript 中直接创建对象来实现此目的。

具体到 Flash 中，new 运算符还可用于创建已在库中定义、但没有放在舞台上的影片剪辑元件的实例。有关此项内容的详细信息，请参阅第 340 页的“[使用 ActionScript 创建 MovieClip 对象](#)”。

常用编程元素

除了声明变量、创建对象实例以及使用属性和方法来处理对象之外，还可以使用其它几个构造块来创建 ActionScript 程序。

运算符

“运算符”是用于执行计算的特殊符号（有时候是词）。这些运算符主要用于数学运算，有时也用于值的比较。通常，运算符使用一个或多个值并“算出”一个结果。例如：

- 加法运算符 (+) 将两个值相加，结果是一个数字：

```
var sum:Number = 23 + 32;
```

- 乘法运算符 (*) 将一个值与另一个值相乘，结果是一个数字：

```
var energy:Number = mass * speedOfLight * speedOfLight;
```

- 等于运算符 (==) 比较两个值以确定是否相等，结果是一个 true 或 false（布尔）值：

```
if (dayOfWeek == "Wednesday")
{
    takeOutTrash();
}
```

如上所示，等于运算符和其它“比较”运算符通常用于 if 语句，以确定是否应执行某些指令。

有关使用运算符的更多详细信息和示例，请参阅第 56 页的“[运算符](#)”。

注释

在编写 ActionScript 时，您通常会希望给自己留一些注释，这些注释可能解释某些代码行如何工作或者为什么做出特定的选择。“代码注释”是一个工具，用于编写计算机应在代码中忽略的文本。ActionScript 包括两种注释：

- 单行注释：在一行中的任意位置放置两个斜杠来指定单行注释。计算机将忽略斜杠后直到该行末尾的所有内容：

```
// This is a comment; it's ignored by the computer.
var age:Number = 10; // Set the age to 10 by default.
```

- 多行注释：多行注释包括一个开始注释标记 /*、注释内容和一个结束注释标记 */。无论注释跨多少行，计算机都将忽略开始标记与结束标记之间的所有内容：

```
/*
This might be a really long description, perhaps describing what
a particular function is used for or explaining a section of code.

In any case, these lines are all ignored by the computer.
*/
```

注释的另一种常见用法是临时“禁用”一行或多行代码 — 例如，如果您要测试执行某项操作的其它方法，或要查明为什么某些 ActionScript 代码没有按您期望的方式工作。

流控制

在程序中，经常需要重复某些动作，仅执行某些动作而不执行其它动作，或根据某些条件执行替代动作等等。“流控制”就是用于控制执行哪些动作。ActionScript 中提供了几种类型的流控制元素。

- 函数：函数类似于快捷方式，提供了一种将一系列操作组合到单个名称下的方法，并可用于执行计算。函数对于处理事件尤为重要，但也可用作组合一系列指令的通用工具。有关函数的详细信息，请参阅第 65 页的“[函数](#)”。
- 循环：使用循环结构，可指定计算机反复执行一组指令，直到达到设定的次数或某些条件改变为止。通常借助循环并使用一个其值在计算机每执行完一次循环后就改变的变量来处理几个相关项。有关循环的详细信息，请参阅第 63 页的“[循环](#)”。
- 条件语句：条件语句提供一种方法，用于指定仅在某些情况下才执行的某些指令或针对不同的条件提供不同的指令集。最常见的一类条件语句是 if 语句。if 语句检查该语句括号中的值或表达式。如果值为 true，则执行大括号中的代码行；否则，将忽略它们。例如：

```
if (age < 20)
{
    // show special teenager-targeted content
}
```

同时使用 if 语句与 else 语句可以指定在条件不为 true 时执行的替代指令：

```
if (username == "admin")
{
    // do some administrator-only things, like showing extra options
}
else
{
    // do some non-administrator things
}
```

有关条件语句的详细信息，请参阅第 61 页的“[条件语句](#)”。

示例：动画公文包片段

该示例的设计目的是让您在第一时机看到如何可以将各段 ActionScript 合并为一个完整的应用程序（如果对于 ActionScript 来说并不过于庞大）。该动画公文包片段是一个示例，演示如何利用现有的线性动画（例如，为客户创建的片段），并添加一些适用于将动画融入在线公文包中的微小的交互式元素。要添加到动画中的交互行为将包括两个查看者可以单击的按钮：一个用于启动动画，另一个用于导航到单独的 URL（例如公文包菜单或创作者的主页）。

创建该片段的过程可以分为四个主要部分：

- 1 准备 FLA 文件以便添加 ActionScript 和交互式元素。
- 2 创建和添加按钮。
- 3 编写 ActionScript 代码。
- 4 测试应用程序。

准备添加交互

向动画添加交互式元素之前，创建一些用于添加新内容的位置将有助于创建 FLA 文件。这包括在舞台上创建可以在其中放置按钮的实际空间，还包括在 FLA 文件中创建“空间”以使不同的项目保持独立。

要创建 **FLA** 以添加交互式元素，请执行下列操作：

- 1 如果还没有要向其中添加交互性的线性动画，请创建一个具有简单动画（例如一个补间动画或补间形状）的新的 FLA 文件。否则，请打开包含您要在本项目中展示的动画的 FLA 文件，并用新名称保存该文件，以创建一个新的工作文件。
- 2 决定您希望两个按钮（一个用于启动动画，另一个用于链接到创造者公文包或主页）在屏幕上显示的位置。如果需要，在舞台上为该新内容清除或添加一些空间。如果该动画还没有启动画面，您可能需要在第 1 帧中创建一个（您可能需要移动动画，使它在第 2 帧或之后的帧中启动）。
- 3 在时间轴中的其它图层上创建一个新图层，并将其重命名为 **buttons**。这将是要在其中添加按钮的图层。
- 4 在 **buttons** 图层之上创建一个新图层，并将其命名为 **actions**。这将是要在其中向应用程序添加 ActionScript 代码的图层。

创建和添加按钮

接下来，将需要真正创建和摆放构成交互式应用程序中心的按钮。

要创建按钮并将其添加到 **FLA**，请执行下列操作：

- 1 使用绘图工具在 **buttons** 图层上创建第一个按钮（“play”按钮）的可视外观，例如，您可以绘制一个水平椭圆，且椭圆顶部有一些文本。
- 2 使用“选择”工具选择单个按钮的所有图形部分。
- 3 在主菜单中，选择“修改”>“转换为元件”。
- 4 在对话框中，选择“按钮”作为元件类型，为该元件赋予一个名称，然后单击“确定”。
- 5 选择按钮之后，在“属性”检查器中为按钮赋予实例名称 **playButton**。
- 6 重复步骤 1 至 5，创建可将查看者转至创作者主页的按钮。将该按钮命名为 **homeButton**。

编写代码

虽然该应用程序的 ActionScript 代码都是在同一个位置输入的，但是该代码可分为三组功能。该代码需要执行的三个任务为：

- 一旦 SWF 文件开始加载（当播放头进入第 1 帧时），就停止播放头。

- 倾听一个事件，该事件在用户单击播放按钮时开始播放 SWF 文件。
- 倾听一个事件，该事件在用户单击创作者主页按钮时将浏览器定向至相应的 URL。

要创建代码，使得在播放头进入第 1 帧时停止播放头，请执行下列操作：

- 1 在 actions 图层的第 1 帧上选择关键帧。
- 2 若要打开“动作”面板，请从主菜单中选择“窗口”>“动作”。
- 3 在“脚本”窗格中，输入以下代码：

```
stop();
```

要编写代码，使得单击播放按钮时启动动画，请执行下列操作：

- 1 在前面步骤中输入的代码的末尾添加两个空行。
- 2 在脚本底部输入以下代码：

```
function startMovie(event:MouseEvent):void
{
    this.play();
}
```

此代码定义一个名为 startMovie() 的函数。调用 startMovie() 时，该函数会导致主时间轴开始播放。

- 3 在上一步中添加的代码的下一行中，输入以下代码行：

```
playButton.addEventListener(MouseEvent.CLICK, startMovie);
```

此代码行将 startMovie() 函数注册为 playButton 的 click 事件的侦听器。也就是说，它使得只要单击名为 playButton 的按钮，就会调用 startMovie() 函数。

要编写代码，使得单击主页按钮时将浏览器定向至某一个 URL，请执行下列操作：

- 1 在前面步骤中输入的代码的末尾添加两个空行。
- 2 在脚本底部输入以下代码：

```
function gotoAuthorPage(event:MouseEvent):void
{
    var targetURL:URLRequest = new URLRequest("http://example.com/");
    navigateToURL(targetURL);
}
```

此代码定义一个名为 gotoAuthorPage() 的函数。此函数首先创建一个表示 URL http://example.com/ 的 URLRequest 实例，然后将该 URL 传递给 navigateToURL() 函数，使用户浏览器打开该 URL。

- 3 在上一步中添加的代码的下一行中，输入以下代码行：

```
homeButton.addEventListener(MouseEvent.CLICK, gotoAuthorPage);
```

此代码行将 gotoAuthorPage() 函数注册为 homeButton 的 click 事件的侦听器。也就是说，它使得只要单击名为 homeButton 的按钮，就会调用 gotoAuthorPage() 函数。

测试应用程序

此时，该应用程序应完全可以工作了。让我们测试一下应用程序，看看到底是不是这样。

要测试该应用程序，请执行下列操作：

- 1 从主菜单中，选择“控制”>“测试影片”。Flash 将创建 SWF 文件，并在 Flash Player 窗口中打开该文件。
- 2 试用这两个按钮，确保它们按您预期的方式工作。

3 如果按钮不起作用，请检查下列事项：

- 这两个按钮是否具有不同的实例名？
- addEventListener() 方法调用使用的名称是否与按钮的实例名相同？
- addEventListener() 方法调用中使用的事件名称是否正确？
- 为各个函数指定的参数是否正确？（两个函数都应具有一个数据类型为 MouseEvent 的参数。）

所有这些错误和大多数其它可能的错误都应在您选择“测试影片”命令或单击按钮时给出错误消息。在“编译器错误”面板中查看编译器错误（这些错误在您第一次选择“测试影片”时发生），并在“输出”面板中查看运行时错误（当 SWF 正在播放时，单击按钮等操作会导致发生这些错误）。

使用 ActionScript 构建应用程序

编写 ActionScript 来构建应用程序的过程不仅仅需要了解使用的语法和类名称。虽然本手册中的大部分信息都围绕这两个主题（语法和使用 ActionScript 类），但是您还需要了解其它一些信息，例如，哪些程序可用于编写 ActionScript、如何组织 ActionScript 代码并将其包括在应用程序中、以及在开发 ActionScript 应用程序时应遵循哪些步骤。

用于组织代码的选项

您可以使用 ActionScript 3.0 代码来实现任何目的，从简单的图形动画到复杂的客户端 - 服务器事务处理系统都可以通过它来实现。您可能希望使用一种或多种不同的方法在项目中包括 ActionScript，具体取决于要构建的应用程序类型。

将代码存储在 Flash 时间轴中的帧中

在 Flash 创作环境中，可以向时间轴中的任何帧添加 ActionScript 代码。该代码将在影片播放期间播放头进入该帧时执行。

通过将 ActionScript 代码放在帧中，可以方便地向使用 Flash 创作工具构建的应用程序添加行为。您可以将代码添加到主时间轴中的任何帧，或任何 MovieClip 元件的时间轴中的任何帧。但是，这种灵活性也有一定的代价。构建较大的应用程序时，这会容易导致无法跟踪哪些帧包含哪些脚本。这会使得随着时间的过去，应用程序越来越难以维护。

许多开发人员将代码仅仅放在时间轴的第 1 帧中，或放在 Flash 文档中的特定图层上，以简化在 Flash 创作工具中组织其 ActionScript 代码的工作。这样，就可以容易地在 Flash FLA 文件中查找和维护代码。但是，要在另一个 Flash 项目中使用相同的代码，您必须将代码复制并粘贴到新文件中。

如果想要以后能够在其它 Flash 项目中使用您的 ActionScript 代码，您需要将代码存储在外部 ActionScript 文件（扩展名为 .as 的文本文件）中。

将代码存储在 ActionScript 文件中

如果您的项目中包括重要的 ActionScript 代码，则最好在单独的 ActionScript 源文件（扩展名为 .as 的文本文件）中组织这些代码。可以采用以下两种方式之一来设置 ActionScript 文件的结构，具体取决于您打算如何在应用程序中使用该文件。

- 非结构化 ActionScript 代码：编写 ActionScript 代码行（包括语句或函数定义），就好像它们是直接在时间轴脚本、MXML 文件等文件中输入的一样。

使用 ActionScript 中的 include 语句或 Adobe Flex MXML 中的 <mx:Script> 标签，可以访问以此方式编写的 ActionScript。ActionScript include 语句会导致在特定位置以及脚本中的指定范围内插入外部 ActionScript 文件的内容，就好像它们是直接在那里输入的一样。在 Flex MXML 语言中，可使用 <mx:Script> 标签来指定源属性，从而标识要在应用程序中的该点处加载的外部 ActionScript 文件。例如，以下标签将加载名为 Box.as 的外部 ActionScript 文件：

```
<mx:Script source="Box.as" />
```

- ActionScript 类定义：定义一个 ActionScript 类，包含它的方法和属性。

定义一个类后，您就可以像对任何内置的 ActionScript 类所做的那样，通过创建该类的一个实例并使用它的属性、方法和事件来访问该类中的 ActionScript 代码。这要求做到下面两点：

- 使用 import 语句来指定该类的全名，以便 ActionScript 编译器知道可以在哪里找到它。例如，如果您希望使用 ActionScript 中的 MovieClip 类，首先需要使用其全名（包括包和类）来导入该类：

```
import flash.display.MovieClip;
```

或者，您也可以导入包含该 MovieClip 类的包，这与针对该包中的每个类编写单独的 import 语句是等效的：

```
import flash.display.*;
```

此规则的唯一例外是，如果在代码中引用的类为顶级类，没有在包中定义，则必须导入该类。

注：在 Flash 中，将自动为附加到时间轴上的帧的脚本导入内置类（在 `flash.*` 包中）。然而，在编写您自己的类时，或是在使用 Flash 创作组件（`fl.*` 包）或在 Flex 中工作时，您将需要显式导入类，以便编写创建该类的实例的代码。

- 编写明确引用类名的代码（通常声明一个用该类作为其数据类型的变量，并创建该类的一个实例以便存储在该变量中）。在 ActionScript 代码中引用其它类名即通知编译器加载该类的定义。例如，如果给定一个称为 Box 的外部类，下面的语句将导致创建 Box 类的一个新实例：

```
var smallBox:Box = new Box(10,20);
```

当编译器第一次遇到对 Box 类的引用时，它会搜索加载的源代码以找到 Box 类的定义。

选择合适的工具

根据项目需求和可用资源，您可能希望使用几个工具中的一个（或结合使用多个工具）来编写和编辑 ActionScript 代码。

Flash 创作工具

除了创建图形和动画的功能之外，Adobe Flash CS4 Professional 还包括处理 ActionScript 代码（附加到 FLA 文件中的元素的代码，或仅包含 ActionScript 代码的外部文件中的代码）的工具。Flash 创作工具最适合于涉及大量的动画或视频的项目，或者您希望自己创建大部分图形资源的项目，尤其适合于用户交互很少或者具有需要 ActionScript 的功能的项目。如果您希望在一个应用程序中既创建可视资源又编写代码，也可能会选择使用 Flash 创作工具来开发 ActionScript 项目。如果您希望使用预置的用户界面组件，但 SWF 较小或便于设置可视外观是项目的主要考虑因素，那么您也可能会选择使用 Flash 创作工具。

Adobe Flash CS4 Professional 包括两个编写 ActionScript 代码的工具：

- “动作”面板：在 FLA 文件中工作时可用，该面板允许您编写附加到时间轴上的帧的 ActionScript 代码。
- “脚本”窗口：“脚本”窗口是专门用于处理 ActionScript (.as) 代码文件的文本编辑器。

Flex Builder

Adobe Flex Builder 是创建带有 Flex 框架的项目的首选工具。除了可视布局和 MXML 编辑工具之外，Flex Builder 还包括一个功能完备的 ActionScript 编辑器，因此可用于创建 Flex 或仅包含 ActionScript 的项目。Flex 应用程序具有以下几个优点：包含一组内容丰富的预置用户界面控件和灵活的动态布局控件，内置了用于处理外部数据源的机制，以及将外部数据链接到用户界面元素。但由于需要额外的代码来提供这些功能，因此 Flex 应用程序的 SWF 文件可能比较大，并且无法像 Flash 应用程序那样轻松地完全重设外观。

如果希望使用 Flex 创建功能完善、数据驱动且内容丰富的 Internet 应用程序，并在一个工具内编辑 ActionScript 代码，编辑 MXML 代码，直观地设置应用程序布局，则应使用 Flex Builder。

第三方 ActionScript 编辑器

由于 ActionScript (.as) 文件存储为简单的文本文件，因此任何能够编辑纯文本文件的程序都可以用来编写 ActionScript 文件。除了 Adobe 的 ActionScript 产品之外，还有几个拥有特定于 ActionScript 的功能的第三方文本编辑程序。您可以使用任何文本编辑程序来编写 MXML 文件或 ActionScript 类。然后，可以使用 Flex SDK（包括 Flex 框架类和 Flex 编译器）来基于这些文件创建 SWF 应用程序（Flex 或仅包含 ActionScript 的应用程序）。或者，很多开发人员也可以使用第三方 ActionScript 编辑器来编写 ActionScript 类，并结合使用 Flash 创作工具来创建图形内容。

在以下情况下，您可以选择使用第三方 ActionScript 编辑器：

- 您希望在单独的程序中编写 ActionScript 代码，而在 Flash 中设计可视元素。
- 将某个应用程序用于非 ActionScript 编程（例如，创建 HTML 页或以其它编程语言构建应用程序），并希望将该应用程序也用于 ActionScript 编码。
- 您希望使用 Flex SDK 而不用 Flash 和 Flex Builder 来创建仅包含 ActionScript 的项目或 Flex 项目。

有一些提供特定于 ActionScript 的支持的代码编辑器值得注意，其中包括：

- [Adobe Dreamweaver® CS4](#)
- [ASDT](#)
- [FDT](#)
- [FlashDevelop](#)
- [PrimalScript](#)
- [SE|PY](#)

ActionScript 开发过程

无论 ActionScript 项目是大还是小，遵循一个过程来设计和开发应用程序都有助于您提高工作效率。下面几个步骤说明了构建使用 ActionScript 3.0 的应用程序的基本开发过程：

1 设计应用程序。

您应先以某种方式描述应用程序，然后再开始构建该应用程序。

2 编写 ActionScript 3.0 代码。

您可以使用 Flash、Flex Builder、Dreamweaver 或文本编辑器来创建 ActionScript 代码。

3 创建 Flash 或 Flex 应用程序文件来运行代码。

在 Flash 创作工具中，此步骤包括：创建新的 FLA 文件、设置发布设置、向应用程序添加用户界面组件以及引用 ActionScript 代码。在 Flex 开发环境中，创建新的应用程序文件涉及：定义该应用程序并使用 MXML 来添加用户界面组件以及引用 ActionScript 代码。

4 发布和测试 ActionScript 应用程序。

这涉及在 Flash 创作环境或 Flex 开发环境中运行应用程序，确保该应用程序执行您期望的所有操作。

请注意，不必按顺序执行这些步骤，或者说不必在完全完成一个步骤后再执行另一步骤。例如，您可能先设计应用程序的一个屏幕（步骤 1），然后创建图形、按钮等等（步骤 3），最后再编写 ActionScript 代码（步骤 2）并进行测试（步骤 4）。您也可以先设计应用程序的一部分，然后再一次添加一个按钮或一个界面元素，并为每个按钮或界面元素编写 ActionScript，并在生成后对之进行测试。虽然记住开发过程的这四个阶段十分有用，但在实际的开发过程中适当地调整各个阶段的顺序通常有助于提高效率。

创建自己的类

创建在项目中使用的类的过程可能令人望而生畏。但是，创建类的过程中更难的部分是设计类，即确定类中将要包含的方法、属性和事件。

类设计策略

面向对象的设计这一主题较为复杂；整个行业的人员都对此学科进行了大量的学术研究和专业实践。尽管如此，下面还是给出了几条建议以帮助您着手进行面向对象的编程。

1 请考虑一下该类的实例将在应用程序中扮演的角色。通常，对象担任以下三种角色之一：

- 值对象：这些对象主要用作数据的容器 — 也就是说，这些对象可能拥有若干个属性和很少的几个方法（有时没有方法）。值对象通常是明确定义的项目的代码表示，例如音乐播放器应用程序中的 **Song** 类（表示单个实际的歌曲）或 **Playlist** 类（表示概念上的一组歌曲）。
- 显示对象：它们是实际显示在屏幕上的对象。例如，用户界面元素（如下拉列表或状态显示）和图形元素（如视频游戏中的角色）等等就是显示对象。
- 应用程序结构：这些对象在应用程序执行的逻辑或处理方面扮演着广泛的支持角色。例如，在仿生学中执行某些计算的对象；在音乐播放器应用程序中负责刻度盘控件与音量显示之间的值同步的对象；管理视频游戏中的规则的对象；或者在绘画应用程序中加载保存的图片的对象。

2 确定类所需的具体功能。不同类型的功能通常会成为类的方法。

3 如果打算将类用作值对象，请确定实例将要包含的数据。这些项是很好的候选属性。

4 由于类是专门为项目而设计的，因此最重要的是提供应用程序所需的功能。回答下列问题可能会对您很有帮助：

- 应用程序将存储、跟踪和处理哪些信息？确定这些信息有助于您识别可能需要的值对象和属性。
- 需要执行的哪些操作 — 例如，在应用程序首次加载时、在单击特定的按钮时、在影片停止播放时以及其它时候需要执行哪些操作？这些是很好的候选方法（如果“动作”仅涉及更改单个值，则是很好的候选属性）。
- 对于任何给定的动作，要执行该动作，该类需要了解哪些信息？这些信息将成为方法的参数。
- 随着应用程序开始工作，应用程序的其它部分需要了解类中的哪些内容将发生更改？这些是很好的候选事件。

5 如果有一个现有的对象类似于您需要的对象，只是缺少您想要添加的一些额外功能，请考虑创建一个子类（在现有类的功能的基础之上构建的类，而不是定义该类自己的所有功能）。例如，如果您希望创建一个将作为屏幕上的可视对象的类，可将一个现有显示对象（如 **Sprite** 或 **MovieClip**）的行为用作该类的基础。在这种情况下，**MovieClip**（或 **Sprite**）是基类，而您的类是该类的扩展。有关创建子类的详细信息，请参阅第 89 页的“[继承](#)”。

编写类的代码

一旦制订了类的设计计划，或至少对该类需要跟踪哪些信息以及该类需要执行哪些动作有了一定了解后，编写类的实际语法就变得非常简单了。

下面是创建自己的 ActionScript 类的最基本步骤：

- 1 在特定于 ActionScript 的程序（如 Flex Builder 或 Flash）、通用编程工具（如 Dreamweaver）或者可用来处理纯文本文档的任何程序中打开一个新的文本文档。
- 2 输入 class 语句定义类的名称。为此，输入单词 **public class**，然后输入类名，后跟一个左大括号和一个右大括号，两个括号之间将是类的内容（方法和属性定义）。例如：

```
public class MyClass
{
}
```

单词 **public** 表示可以从任何其它代码中访问该类。有关其它选项，请参阅第 78 页的“[访问控制命名空间属性](#)”。

- 3 键入 package 语句以指示包含该类的包的名称。语法是单词 package，后跟完整的包名称，再跟左大括号和右大括号（括号之间将是 class 语句块）。例如，我们将上一步中的代码更改为：

```
package mypackage
{
    public class MyClass
    {
    }
}
```

- 4 在类体中使用 var 语句定义该类中的每个属性；语法与用于声明任何变量的语法相同（并增加了 public 修饰符）。例如，在类定义的左大括号与右大括号之间添加下列行将创建名为 textVariable、numericVariable 和 dateVariable 属性：

```
public var textVariable:String = "some default value";
public var numericVariable:Number = 17;
public var dateVariable:Date;
```

- 5 使用与函数定义所用的相同语法来定义类中的每个方法。例如：

- 要创建 myMethod() 方法，应输入：

```
public function myMethod(param1:String, param2:Number) :void
{
    // do something with parameters
}
```

- 要创建一个构造函数（在创建类实例的过程中调用的特殊方法），应创建一个名称与类名称完全匹配的方法：

```
public function MyClass()
{
    // do stuff to set initial values for properties
    // and otherwise set up the object
    textVariable = "Hello there!";
    dateVariable = new Date(2001, 5, 11);
}
```

如果没有在类中包括构造函数方法，则编译器将会在类中自动创建一个空构造函数（没有参数和语句）。

您还可以定义其它几个类元素。这些元素更为复杂。

- “存取器”是方法与属性之间的一个特殊交点。在编写代码来定义类时，可以像编写方法一样来编写存取器，这样就可以执行多个动作（而不是像在定义属性时那样，只能读取值或赋值）。但是，在创建类的实例时，可将存取器视为属性——仅使用名称来读取值或赋值。有关详细信息，请参阅第 83 页的“[get 和 set 存取器方法](#)”。
- ActionScript 中的事件不是使用特定语法来定义的。应使用 EventDispatcher 类的功能来定义类中的事件，以便跟踪事件监听器并将事件通知给它们。有关在您自己的类中创建事件的详细信息，请参阅第 207 页的“[处理事件](#)”。

示例：创建基本应用程序

可以使用 Flash、Flex Builder、Dreamweaver 或任何文本编辑器来创建一个扩展名为 .as 的外部 ActionScript 源文件。

ActionScript 3.0 可以用在许多应用程序开发环境（包括 Flash 创作工具和 Flex Builder 工具）中。

本节将引导您完成使用 Flash 创作工具或 Flex Builder 创建和增强一个简单的 ActionScript 3.0 应用程序的步骤。您构建的应用程序会呈现出一种在 Flash 和 Flex 应用程序中使用外部 ActionScript 3.0 类文件的简单模式。在本手册中该模式将会应用于其它所有应用程序范例。

设计 ActionScript 应用程序

您应先对要构建的应用程序有一些想法，然后再开始构建该应用程序。

设计的表示可以像应用程序的名称和应用程序用途的简要说明一样简单，也可以像一组包含许多统一建模语言 (UML) 图示的需求文档一样复杂。本手册不会详细讨论软件设计学科，但希望提醒读者记住应用程序设计是开发 ActionScript 应用程序的一个重要步骤。

我们的第一个 ActionScript 应用程序示例是一个标准的“Hello World”应用程序，该应用程序的设计非常简单：

- 该应用程序将被称为 **HelloWorld**。
- 这个应用程序将会显示一个包含“Hello World!”字样的文本字段。
- 为了便于重用，该应用程序将使用一个名为 **Greeter** 的面向对象的类，该类既可以在 Flash 文档中使用，也可以在 Flex 应用程序中使用。
- 在创建该应用程序的一个基本版本之后，您将添加新功能，让用户输入一个用户名并让应用程序对照已知用户列表来检查该用户名。

有了这一简明的定义之后，您可以开始构建该应用程序了。

创建 HelloWorld 项目和 Greeter 类

Hello World 应用程序的设计说明指出该应用程序的代码应易于重用。为了实现此目标，该应用程序使用一个名为 **Greeter** 的面向对象的类，该类可以在 Flex Builder 或 Flash 创作工具创建的应用程序中使用。

要在 Flash 创作工具中创建 **Greeter** 类，请执行下列操作：

- 1 在 Flash 创作工具中，选择“文件”>“新建”。
- 2 在“新建文档”对话框中，选择“ActionScript 文件”，然后单击“确定”。
将显示一个新的 ActionScript 编辑窗口。
- 3 选择“文件”>“保存”。选择一个文件夹以包含您的应用程序，将 ActionScript 文件命名为 **Greeter.as**，然后单击“确定”。
继续完成第 25 页的“[在 Greeter 类中添加代码](#)”。

在 Greeter 类中添加代码

Greeter 类定义一个对象 **Greeter**，您可以在 HelloWorld 应用程序中使用该对象。

要向 **Greeter** 类中添加代码，请执行以下操作：

- 1 将以下代码键入该新文件中：

```
package
{
    public class Greeter
    {
        public function sayHello():String
        {
            var greeting:String;
            greeting = "Hello World!";
            return greeting;
        }
    }
}
```

Greeter 类包含一个 **sayHello()** 方法，该方法返回字符串“Hello World!”。

- 2 选择“文件”>“保存”保存此 ActionScript 文件。

Greeter 类准备就绪，可在应用程序中使用。

创建使用 ActionScript 代码的应用程序

您构建的 Greeter 类定义一组自包含的软件功能，但它不表示完整的应用程序。要使用该类，需要创建一个 Flash 文档或 Flex 应用程序。

HelloWorld 应用程序创建 Greeter 类的一个新实例。下面说明如何将 Greeter 类附加到应用程序。

要使用 Flash 创作工具创建 ActionScript 应用程序，请执行下列操作：

- 1 选择“文件”>“新建”。
- 2 在“新建文档”对话框中，选择“Flash 文档”，然后单击“确定”。
将显示一个新的 Flash 窗口。
- 3 选择“文件”>“保存”。选择包含该 Greeter.as 类文件的同一文件夹，将 Flash 文档命名为 HelloWorld.fla，然后单击“确定”。
- 4 在 Flash 的“工具”调色板中，选择“文本”文件，然后在舞台上拖动以定义一个新的文本字段，该字段宽约 300 像素，高约 100 像素。
- 5 在“属性”面板中，在仍然选中舞台中的文本字段时，将文本类型设置为“动态文本”，并键入 mainText 作为该文本字段的实例名称。
- 6 单击主时间轴的第 1 帧。
- 7 在“动作”面板中键入以下脚本：

```
var myGreeter:Greeter = new Greeter();
mainText.text = myGreeter.sayHello();
```

- 8 保存该文件。

继续完成第 26 页的“[发布和测试 ActionScript 应用程序](#)”。

发布和测试 ActionScript 应用程序

软件开发是一个重复的过程。编写一些代码，尝试编译这些代码，然后编辑代码，直到能够完全编译这些代码为止。运行编译后的应用程序，测试该应用程序以确定应用程序是否实现了预期的设计，如果没有，应再次编辑代码，直到实现预期的设计为止。Flash 和 Flex Builder 开发环境提供了多种发布、测试和调试应用程序的方法。

下面是在每种环境中测试 HelloWorld 应用程序的基本步骤。

要使用 Flash 创作工具发布和测试 ActionScript 应用程序，请执行下列操作：

- 1 发布您的应用程序并观察编译错误。在 Flash 创作工具中，选择“控制”>“测试影片”，编译您的 ActionScript 代码并运行 HelloWorld 应用程序。
- 2 如果测试应用程序时“输出”窗口中显示任何错误或警告，请在 HelloWorld.fla 或 HelloWorld.as 文件中修复导致这些错误的根源，然后重新测试该应用程序。
- 3 如果没有编译错误，您将看到一个显示 Hello World 应用程序的 Flash Player 窗口。

您刚才创建了一个简单但完整的面向对象的应用程序，该应用程序使用 ActionScript 3.0。继续完成第 26 页的“[改进 HelloWorld 应用程序](#)”。

改进 HelloWorld 应用程序

若要使该应用程序更有趣，现在可让应用程序要求用户输入用户名并对照预定义的名称列表来验证该用户名。

首先，更新 Greeter 类以添加新功能。然后更新应用程序以使用新功能。

要更新 **Greeter.as** 文件, 请执行以下操作:

- 1 打开 Greeter.as 文件。
- 2 将文件的内容更改为如下内容 (新行和更改的行以粗体显示) :

```

package
{
    public class Greeter
    {
        /**
         * Defines the names that should receive a proper greeting.
         */
        public static var validNames:Array = ["Sammy", "Frank", "Dean"];

        /**
         * Builds a greeting string using the given name.
         */
        public function sayHello(userName:String = ""):String
        {
            var greeting:String;
            if (userName == "")
            {
                greeting = "Hello. Please type your user name, and then press
                           the Enter key.";
            }
            else if (validName(userName))
            {
                greeting = "Hello, " + userName + ".";
            }
            else
            {
                greeting = "Sorry " + userName + ", you are not on the list.";
            }
            return greeting;
        }

        /**
         * Checks whether a name is in the validNames list.
         */
        public static function validName(inputName:String = ""):Boolean
        {
            if (validNames.indexOf(inputName) > -1)
            {
                return true;
            }
            else
            {
                return false;
            }
        }
    }
}

```

现在, Greeter 类拥有很多新功能:

- validNames 数组列出了有效的用户名。在加载 Greeter 类时该数组将初始化为包含三个名称的列表。
- sayHello() 方法现在接受一个用户名并根据一些条件来更改问候语。如果 userName 是一个空字符串 ("") , 则 greeting 属性将设置为提示用户输入用户名的语句。如果用户名有效, 则问候语会是 "Hello, userName"。最后, 如果这两个条件都不满足, 则 greeting 变量设置为 "Sorry userName, you are not on the list"。

- 如果在 validNames 数组中找到 inputName，则 validName() 方法将返回 true；否则，返回 false。语句 validNames.indexOf(inputName) 对照 inputName 字符串检查 validNames 数组中的每个字符串。Array.indexOf() 方法返回数组中某个对象的第一个实例的索引位置；如果在数组中找不到该对象，则返回值 -1。

接下来您将编辑引用该 ActionScript 类的 Flash 或 Flex 文件。

要使用 **Flash** 创作工具修改该应用程序，请执行下列操作：

- 打开 HelloWorld.fla 文件。
- 修改第 1 帧中的脚本，将一个空字符串 ("") 传递到 Greeter 类的 sayHello() 方法：

```
var myGreeter:Greeter = new Greeter();
mainText.text = myGreeter.sayHello("");
```
- 在“工具”调色板中选择“文本”工具，然后在舞台上创建两个新的文本字段，这两个字段直接并排放置在现有的 mainText 文本字段之下。
- 在第一个新文本字段中，键入文本 **User Name:** 作为标签。
- 选择另一个新文本字段，并在“属性”检查器中选择 **InputText** 作为该文本字段的类型。选择“单行”作为“行类型”。键入 **textIn** 作为实例名。
- 单击主时间轴的第 1 帧。
- 在“动作”面板中，在现有脚本的末尾添加以下行：

```
mainText.border = true;
textIn.border = true;

textIn.addEventListener(KeyboardEvent.KEY_DOWN, keyPressed);

function keyPressed(event:KeyboardEvent):void
{
    if (event.keyCode == Keyboard.ENTER)
    {
        mainText.text = myGreeter.sayHello(textIn.text);
    }
}
```

新代码添加以下功能：

- 前两行只定义两个文本字段的边框。
- 输入文本字段（例如 textIn 字段）具有一组可以调度的事件。使用 addEventListener() 方法可以定义一个函数，该函数在发生某一类型的事件时运行。在本例中，该事件指的是按键盘上的某个键。
- keyPressed() 自定义函数会检查按下的键是否为 Enter 键。如果是，则该函数调用 myGreeter 对象的 sayHello() 方法，同时传递 textIn 文本字段中的文本作为参数。该方法基于传入的值返回字符串 “greeting”。返回的字符串随后分配给 mainText 文本字段的 text 属性。

第 1 帧的完整脚本如下所示：

```
var myGreeter:Greeter = new Greeter();
mainText.text = myGreeter.sayHello("");

mainText.border = true;
textIn.border = true;

textIn.addEventListener(KeyboardEvent.KEY_DOWN, keyPressed);

function keyPressed(event:KeyboardEvent):void
{
    if (event.keyCode == Keyboard.ENTER)
    {
        mainText.text = myGreeter.sayHello(textIn.text);
    }
}
```

8 保存该文件。

9 选择“控制”>“测试影片”，运行应用程序。

运行该应用程序时，将提示您输入用户名。如果用户名有效（Sammy、Frank 或 Dean），应用程序则会显示“hello”确认消息。

运行后续示例

请注意，您已开发并运行了“Hello World”ActionScript 3.0 应用程序，您应掌握了运行本手册中提供的其它代码示例所需的基础知识。

测试本章内的示例代码清单

完成本手册内容的过程中，您可能需要试验用于阐释各个主题的示例代码清单。该测试可能包括显示程序中特定点的变量值，也可能包括查看屏幕上的内容或与该内容交互。为了测试可视内容或交互，将在代码清单之前或代码清单中描述一些必需的元素：您只需要使用所述的元素创建一个文档以测试代码。如果您想要查看程序中某个给定点的变量值，可以通过几种方法来实现。一种方法是使用调试器，例如 Flex Builder 和 Flash 中的内置调试器。但是，对于简单的测试，最简单的方法是只需将变量值输出到某个您可以查看的位置。

下列步骤将帮助您创建一个 Flash 文档，您可以使用该文档来测试代码清单并查看变量值：

要创建用于测试本章内示例的 **Flash** 文档，请执行下列操作：

- 1 创建一个新的 Flash 文档，并将其保存在硬盘驱动器上。
- 2 要在舞台上的文本字段中显示测试值，请激活“文本”工具并在舞台上创建一个新的动态文本字段。最有用的是一个较宽、较高的文本字段，该字段的“线条类型”设置为“多行”，且启用了边框。在“属性”检查器中，为该文本字段赋予一个实例名称（例如“outputText”）。要将值写入该文本字段中，您需要将调用 `appendText()` 方法的代码添加到示例代码中（如下所述）。
- 3 或者，可以将一个 `trace()` 函数调用添加到代码清单中（如下所述）以查看示例的结果。
- 4 要测试给定的示例，请将代码清单复制到“动作”面板中；如果需要，添加一个 `trace()` 函数调用，或使用其 `appendText()` 方法向文本字段添加值。
- 5 从主菜单中，选择“控制”>“测试影片”以创建一个 SWF 文件并查看结果。

由于此方法是用于查看变量的值，所以您可以通过两种途径在试验示例时方便地查看变量值：将值写入舞台上的文本字段实例中，或使用 `trace()` 函数将值输出到“输出”面板。

- `trace()` 函数：ActionScript `trace()` 函数将传递给它的任何参数的值（变量或文本表达式）写入“输出”面板。本手册中的许多示例列表已经包括了一个 `trace()` 函数调用，因此对于这些列表，您将只需要将代码复制到您的文档中并测试项目。如果要使用 `trace()` 测试不包括该方法的代码清单中的变量值，则只需向该代码清单添加一个 `trace()` 调用，同时传递该变量作为参数。例如，如果遇到类似本章中的代码清单，

```
var albumName:String = "Three for the money";
```

您可以将代码复制到“动作”面板，然后添加对 `trace()` 函数的调用，以测试代码清单的结果：

```
var albumName:String = "Three for the money";
trace("albumName =", albumName);
```

运行该程序时，将输出以下行：

```
albumName = Three for the money
```

每个 `trace()` 函数调用可以采用多个参数，这些参数排列在一起，组成一个输出行。每个 `trace()` 函数调用的末尾添加了换行符，因此各个 `trace()` 调用将在单独的行中输出。

- 舞台上的文本字段：如果您不想使用 `trace()` 函数，可以使用“文本”工具向舞台添加一个动态文本字段，将值写出到该文本字段中，以查看代码清单的结果。可以使用 `TextField` 类的 `appendText()` 方法在该文本字段内容的末尾添加一个字符串值。要使用 ActionScript 访问该文本字段，必须在“属性”检查器中为其赋予一个实例名。例如，如果您的文本字段具有实例名 `outputText`，则可以使用以下代码来查看变量 `albumName` 的值：

```
var albumName:String = "Three for the money";
outputText.appendText("albumName = ");
outputText.appendText(albumName);
```

此代码将下面的文本写入名为 `outputText` 的文本字段中：

```
albumName = Three for the money
```

如该示例所示，`appendText()` 方法将文本添加到与前面的内容相同的行中，因此，可以使用多个 `appendText()` 调用向同一文本行中添加多个值。要强制将文本写入下一行，您可以追加一个换行符 ("\\n")：

```
outputText.appendText("\\n"); // adds a line break to the text field
```

与 `trace()` 函数不同，`appendText()` 方法仅接受一个值作为参数。该值必须是字符串（字符串实例或字符串文本）。要输出非字符串变量的值，必须首先将值转换为字符串。最简单的方法是调用对象的 `toString()` 方法：

```
var albumYear:int = 1999;
outputText.appendText("albumYear = ");
outputText.appendText(albumYear.toString());
```

处理章节结尾的示例

与本章一样，本手册中的大多数章节的结尾都包含一个重要示例，它将该章节中讨论的很多内容串联在一起。但是，与本章中的 **Hello World** 示例不同，这些示例不是以分步教程的形式提供的。每个示例中相关的 ActionScript 3.0 代码将突出显示并加以讨论，但是不会提供在特定开发环境中运行示例的说明。但是，与本手册一起发布的示例文件将包括在所选择的开发环境中轻松编译和运行示例所需的全部文件。

第 4 章：ActionScript 语言及其语法

ActionScript 3.0 既包含 ActionScript 核心语言又包含 Adobe Flash Player 应用程序编程接口 (API)。核心语言是定义语言语法以及顶级数据类型的 ActionScript 部分。ActionScript 3.0 提供对 Flash Player 的编程访问。

本章简要介绍 ActionScript 核心语言及其语法。在阅读完本章之后，您应能够对如何处理数据类型和变量、如何使用正确的语法以及如何控制程序中的数据流等方面有一个基本的了解。

语言概述

对象是 ActionScript 3.0 语言的核心 — 它们是 ActionScript 3.0 语言的基本构造块。您所声明的每个变量、所编写的每个函数以及所创建的每个类实例都是一个对象。可以将 ActionScript 3.0 程序视为一组执行任务、响应事件以及相互通信的对象。

熟悉 Java 或 C++ 中面向对象的编程 (OOP) 的程序员可能会将对象视为包含以下两类成员的模块：存储在成员变量或属性中的数据，以及可通过方法访问的行为。ActionScript 3.0 采用一种相似但稍有不同的方式定义对象。在 ActionScript 3.0 中，对象只是属性的集合。这些属性是一些容器，除了保存数据，还保存函数或其它对象。以这种方式附加到对象的函数称为方法。

尽管具有 Java 或 C++ 背景的程序员可能会觉得 ActionScript 3.0 中的定义有些奇怪，但实际上，用 ActionScript 3.0 类定义对象类型与在 Java 或 C++ 中定义类的方式非常相似。在讨论 ActionScript 对象模型和其它高级主题时，了解这两种对象定义之间的区别很重要，但在其它大多数情况下，术语属性表示类成员变量（而不是方法）。例如，《ActionScript 3.0 语言和组件参考》使用术语属性表示变量或 **getter-setter** 属性。该手册使用术语方法表示作为类的一部分的函数。

ActionScript 中的类与 Java 或 C++ 中的类之间有一个细小的差别，那就是 ActionScript 中的类不仅仅是抽象实体。ActionScript 类由存储类的属性和方法的类对象表示。这样便可以使用可能不为 Java 和 C++ 程序员所熟悉的方法，例如，在类或包的顶级包括语句或可执行代码。

ActionScript 类与 Java 或 C++ 类之间还有一个区别，那就是每个 ActionScript 类都有一个原型对象。在早期版本的 ActionScript 中，原型对象链接成原型链，共同作为整个类继承层次结构的基础。但是，在 ActionScript 3.0 中，原型对象在继承系统中仅扮演很小的角色。但是，原型对象仍非常有用，如果您希望在类的所有实例中共享某个属性及其值，可以使用原型对象来代替静态属性和方法。

过去，高级 ActionScript 程序员可以用特殊的内置语言元素来直接操作原型链。现在，由于 ActionScript 语言为基于类的编程接口提供了更成熟的实现，因此其中的许多特殊语言元素（如 `__proto__` 和 `__resolve`）不再是该语言的一部分。而且，内部继承机制的优化预先排除了对继承机制的直接访问，从而大大改善了 Flash Player 和 Adobe AIR 的性能。

对象和类

在 ActionScript 3.0 中，每个对象都是由类定义的。可将类视为某一类对象的模板或蓝图。类定义中可以包括变量和常量以及方法，前者用于保存数据值，后者是封装绑定到类的行为的函数。存储在属性中的值可以是基元值，也可以是其它对象。基元值是指数字、字符串或布尔值。

ActionScript 中包含许多属于核心语言的内置类。其中的某些内置类（如 `Number`、`Boolean` 和 `String`）表示 ActionScript 中可用的基元值。其它类（如 `Array`、`Math` 和 `XML` 类）定义更加复杂的对象。

所有的类（无论是内置类还是用户定义的类）都是从 `Object` 类派生的。以前在 ActionScript 方面有经验的程序员一定要注意到，`Object` 数据类型不再是默认的数据类型，尽管其它所有类仍从它派生。在 ActionScript 2.0 中，下面的两行代码等效，因为缺乏类型注释意味着变量为 `Object` 类型：

```
var someObj:Object;
var someObj;
```

但是，ActionScript 3.0 引入了无类型变量这一概念，这一类变量可通过以下两种方法来指定：

```
var someObj:.*;
var someObj;
```

无类型变量与 Object 类型的变量不同。二者的主要区别在于无类型变量可以保存特殊值 undefined，而 Object 类型的变量则不能保存该值。

您可以使用 class 关键字来定义自己的类。在方法声明中，可通过以下三种方法来声明类属性 (property)：用 const 关键字定义常量，用 var 关键字定义变量，用 get 和 set 属性 (attribute) 定义 getter 和 setter 属性 (property)。可以用 function 关键字来声明方法。

可使用 new 运算符来创建类的实例。下面的示例创建 Date 类的一个名为 myBirthday 的实例。

```
var myBirthday:Date = new Date();
```

包和命名空间

包和命名空间是两个相关的概念。使用包，可以通过有利于共享代码并尽可能减少命名冲突的方式将多个类定义捆绑在一起。使用命名空间，可以控制标识符（如属性名和方法名）的可见性。无论命名空间位于包的内部还是外部，都可以应用于代码。包可用于组织类文件，命名空间可用于管理各个属性和方法的可见性。

包

在 ActionScript 3.0 中，包是用命名空间实现的，但包和命名空间并不同义。在声明包时，可以隐式创建一个特殊类型的命名空间并保证它在编译时是已知的。显式创建的命名空间在编译时不必是已知的。

下面的示例使用 package 指令来创建一个包含单个类的简单包：

```
package samples
{
    public class SampleCode
    {
        public var sampleGreeting:String;
        public function sampleFunction()
        {
            trace(sampleGreeting + " from sampleFunction()");
        }
    }
}
```

在本例中，该类的名称是 SampleCode。由于该类位于 samples 包中，因此编译器在编译时会自动将其类名称限定为完全限定名称：samples.SampleCode。编译器还限定任何属性或方法的名称，以便 sampleGreeting 和 sampleFunction() 分别变成 samples.SampleCode.sampleGreeting 和 samples.SampleCode.sampleFunction()。

许多开发人员（尤其是那些具有 Java 编程背景的人）可能会选择只将类放在包的顶级。但是，ActionScript 3.0 不但支持将类放在包的顶级，而且还支持将变量、函数甚至语句放在包的顶级。此功能的一个高级用法是，在包的顶级定义一个命名空间，以便它对于该包中的所有类均可用。但是，请注意，在包的顶级只允许使用两个访问说明符：public 和 internal。Java 允许将嵌套类声明为私有，而 ActionScript 3.0 则不同，它既不支持嵌套类也不支持私有类。

但是，在其它许多方面，ActionScript 3.0 中的包与 Java 编程语言中的包非常相似。从上一个示例可看出，完全限定的包引用点运算符（.）来表示，这与 Java 相同。可以用包将代码组织成直观的分层结构，以供其他程序员使用。这样，您就可以将自己所创建的包与他人共享，还可以在自己的代码中使用他人创建的包，从而推动了代码共享。

使用包还有助于确保所使用的标识符名称是唯一的，而且不与其它标识符名称冲突。事实上，有些人认为这才是包的主要优点。例如，假设两个希望相互共享代码的程序员各创建了一个名为 **SampleCode** 的类。如果没有包，这样就会造成名称冲突，唯一的解决方法就是重命名其中的一个类。但是，使用包，就可以将其中的一个（最好是两个）类放在具有唯一名称的包中，从而轻松地避免了名称冲突。

您还可以在包名称中嵌入点来创建嵌套包，这样就可以创建包的分层结构。ActionScript 3.0 提供的 **flash.xml** 包就是一个很好的示例。**flash.xml** 包嵌套在 **flash** 包中。

flash.xml 包中包含在早期的 ActionScript 版本中使用的旧 XML 分析器。该分析器现在之所以包含在 **flash.xml** 包中，原因之一是，旧 XML 类的名称与一个新 XML 类的名称冲突，这个新 XML 类实现 ActionScript 3.0 中的 XML for ECMAScript (E4X) 规范功能。

尽管首先将旧的 XML 类移入包中是一个不错的主意，但是旧 XML 类的大多数用户都会导入 **flash.xml** 包，这样，除非您总是记得使用旧 XML 类的完全限定名称 (**flash.xml.XML**)，否则同样会造成名称冲突。为避免这种情况，现在已将旧 XML 类命名为 **XMLDocument**，如下面的示例所示：

```
package flash.xml
{
    class XMLDocument {}
    class XMLNode {}
    class XMLSocket {}
}
```

大部分 ActionScript 3.0 都划分到 **flash** 包中。例如，**flash.display** 包中包含显示列表 API，**flash.events** 包中包含新的事件模型。

创建包

ActionScript 3.0 在包、类和源文件的组织方式上具有很大的灵活性。早期的 ActionScript 版本只允许每个源文件有一个类，而且要求源文件的名称与类名称匹配。ActionScript 3.0 允许在一个源文件中包括多个类，但是，每个文件中只有一个类可供该文件外部的代码使用。换言之，每个文件中只有一个类可以在包声明中进行声明。您必须在包定义的外部声明其它任何类，以使这些类对于该源文件外部的代码不可见。在包定义内部声明的类的名称必须与源文件的名称匹配。

ActionScript 3.0 在包的声明方式上也具有更大的灵活性。在早期的 ActionScript 版本中，包只是表示用于放置源文件的目录，您不必用 **package** 语句来声明包，而是在类声明中将包名称包括在完全限定的类名称中。在 ActionScript 3.0 中，尽管包仍表示目录，但是它现在不只包含类。在 ActionScript 3.0 中，使用 **package** 语句来声明包，这意味着您还可以在包的顶级声明变量、函数和命名空间，甚至还可以在包的顶级包括可执行语句。如果在包的顶级声明变量、函数或命名空间，则在顶级只能使用 **public** 和 **internal** 属性，并且每个文件中只能有一个包级声明使用 **public** 属性（无论该声明是类声明、变量声明、函数声明还是命名空间声明）。

包的作用是组织代码并防止名称冲突。您不应将包的概念与类继承这一不相关的概念混淆。位于同一个包中的两个类具有共同的命名空间，但是它们在其它任何方面都不必相关。同样，在语义方面，嵌套包可以与其父包无关。

导入包

如果您希望使用位于某个包内部的特定类，则必须导入该包或该类。这与 ActionScript 2.0 不同，在 ActionScript 2.0 中，类的导入是可选的。

以本章前面的 **SampleCode** 类示例为例。如果该类位于名为 **samples** 的包中，那么，在使用 **SampleCode** 类之前，您必须使用下列导入语句之一：

```
import samples.*;
```

或

```
import samples.SampleCode;
```

通常，`import` 语句越具体越好。如果您只打算使用 `samples` 包中的 `SampleCode` 类，则应只导入 `SampleCode` 类，而不应导入该类所属的整个包。导入整个包可能会导致意外的名称冲突。

还必须将定义包或类的源代码放在类路径 内部。类路径是用户定义的本地目录路径列表，它决定了编译器将在何处搜索导入的包和类。类路径有时称为生成路径 或源路径。

在正确地导入类或包之后，可以使用类的完全限定名称 (`samples.SampleCode`)，也可以只使用类名称本身 (`SampleCode`)。

当同名的类、方法或属性会导致代码不明确时，完全限定的名称非常有用，但是，如果将它用于所有的标识符，则会使代码变得难以管理。例如，在实例化 `SampleCode` 类的实例时，使用完全限定的名称会导致代码冗长：

```
var mySample:samples.SampleCode = new samples.SampleCode();
```

包的嵌套级别越高，代码的可读性越差。如果您确信不明确的标识符不会导致问题，就可以通过使用简单的标识符来提高代码的可读性。例如，如果在实例化 `SampleCode` 类的新实例时仅使用类标识符，代码就会简短得多。

```
var mySample:SampleCode = new SampleCode();
```

如果您尝试使用标识符名称，而不先导入相应的包或类，编译器将找不到类定义。另一方面，即便您导入了包或类，只要尝试定义的名称与所导入的名称冲突，也会产生错误。

创建包时，该包的所有成员的默认访问说明符是 `internal`，这意味着，默认情况下，包成员仅对其所在包的其它成员可见。如果您希望某个类对包外部的代码可用，则必须将该类声明为 `public`。例如，下面的包包含 `SampleCode` 和 `CodeFormatter` 两个类：

```
// SampleCode.as file
package samples
{
    public class SampleCode {}
}

// CodeFormatter.as file
package samples
{
    class CodeFormatter {}
}
```

`SampleCode` 类在包的外部可见，因为该类声明为 `public` 类。但是，`CodeFormatter` 类仅在 `samples` 包的内部可见。如果您尝试访问位于 `samples` 包外部的 `CodeFormatter` 类，将会产生一个错误，如下面的示例所示：

```
import samples.SampleCode;
import samples.CodeFormatter;
var mySample:SampleCode = new SampleCode(); // okay, public class
var myFormatter:CodeFormatter = new CodeFormatter(); // error
```

如果您希望这两个类在包外部均可用，必须将它们都声明为 `public`。不能将 `public` 属性应用于包声明。

完全限定的名称可用来解决在使用包时可能发生的名称冲突。如果您导入两个包，但它们用同一个标识符来定义类，就可能会发生名称冲突。例如，请考虑下面的包，该包也有一个名为 `SampleCode` 的类：

```
package langref.samples
{
    public class SampleCode {}
}
```

如果按如下方式导入两个类，在引用 `SampleCode` 类时将会发生名称冲突：

```
import samples.SampleCode;
import langref.samples.SampleCode;
var mySample:SampleCode = new SampleCode(); // name conflict
```

编译器无法确定要使用哪个 `SampleCode` 类。要解决此冲突，必须使用每个类的完全限定名称，如下所示：

```
var sample1:samples.SampleCode = new samples.SampleCode();
var sample2:langref.samples.SampleCode = new langref.samples.SampleCode();
```

注：具有 C++ 背景的程序员通常会将 `import` 语句与 `#include` 混淆。`#include` 指令在 C++ 中是必需的，因为 C++ 编译器一次处理一个文件，而且除非显式包括了头文件，否则将不会在其它文件中查找类定义。ActionScript 3.0 有一个 `include` 指令，但是它的作用不是为了导入类和包。要在 ActionScript 3.0 中导入类或包，必须使用 `import` 语句，并将包含该包的源文件放在类路径中。

命名空间

通过命名空间可以控制所创建的属性和方法的可见性。请将 `public`、`private`、`protected` 和 `internal` 访问控制说明符视为内置的命名空间。如果这些预定义的访问控制说明符无法满足您的要求，您可以创建自己的命名空间。

如果您熟悉 XML 命名空间，那么，您对这里讨论的大部分内容不会感到陌生，但是 ActionScript 实现的语法和细节与 XML 的稍有不同。即使您以前从未使用过命名空间，也没有关系，因为命名空间概念本身很简单，但是其实现涉及一些您需要了解的特定术语。

要了解命名空间的工作方式，有必要先了解属性或方法的名称总是包含两部分：标识符和命名空间。标识符通常被视为名称。例如，下面的类定义中的标识符是 `sampleGreeting` 和 `sampleFunction()`：

```
class SampleCode
{
    var sampleGreeting:String;
    function sampleFunction () {
        trace(sampleGreeting + " from sampleFunction()");
    }
}
```

只要定义不以命名空间属性开头，就会用默认 `internal` 命名空间限定其名称，这意味着，这些定义仅对同一个包中的调用方可见。如果编译器设置为严格模式，则编译器会发出一个警告，指明 `internal` 命名空间将应用于没有命名空间属性的任何标识符。为了确保标识符可在任何位置使用，您必须在标识符名称的前面明确加上 `public` 属性。在上面的示例代码中，`sampleGreeting` 和 `sampleFunction()` 都有一个命名空间值 `internal`。

使用命名空间时，应遵循以下三个基本步骤。第一，必须使用 `namespace` 关键字来定义命名空间。例如，下面的代码定义 `version1` 命名空间：

```
namespace version1;
```

第二，在属性或方法声明中，使用命名空间（而非访问控制说明符）来应用命名空间。下面的示例将一个名为 `myFunction()` 的函数放在 `version1` 命名空间中：

```
version1 function myFunction() {}
```

第三，在应用了该命名空间后，可以使用 `use` 指令进行引用，也可以使用一个命名空间来限定标识符的名称。下面的示例通过 `use` 指令来引用 `myFunction()` 函数：

```
use namespace version1;
myFunction();
```

您还可以使用限定名称引用 `myFunction()` 函数，如下面的示例所示：

```
version1::myFunction();
```

定义命名空间

命名空间中包含一个名为统一资源标识符 (URI) 的值，该值有时称为命名空间名称。使用 URI 可确保命名空间定义的唯一性。

可通过使用以下两种方法之一来声明命名空间定义，以创建命名空间：像定义 XML 命名空间那样使用显式 URI 定义命名空间；省略 URI。下面的示例说明如何使用 URI 来定义命名空间：

```
namespace flash_proxy = "http://www.adobe.com/flash/proxy";
```

URI 用作该命名空间的唯一标识字符串。如果您省略 URI（如下面的示例所示），则编译器将创建一个唯一的内部标识字符串来代替 URI。您对于这个内部标识字符串不具有访问权限。

```
namespace flash_proxy;
```

在定义了命名空间（具有 URI 或没有 URI）后，就不能在同一个作用域内重新定义该命名空间。如果尝试定义的命名空间以前在同一个作用域内定义过，则将生成编译器错误。

如果在某个包或类中定义了一个命名空间，则该命名空间可能对于此包或类外部的代码不可见，除非使用了相应的访问控制说明符。例如，下面的代码显示了在 `flash.utils` 包中定义的 `flash_proxy` 命名空间。下面的示例中没有使用访问控制说明符，这意味着 `flash_proxy` 命名空间仅对于 `flash.utils` 包内部的代码可见，而对于该包外部的任何代码都不可见：

```
package flash.utils
{
    namespace flash_proxy;
}
```

下面的代码使用 `public` 属性使 `flash_proxy` 命名空间对该包外部的代码可见：

```
package flash.utils
{
    public namespace flash_proxy;
}
```

应用命名空间

应用命名空间意味着在命名空间中放置定义。可以放在命名空间中的定义包括函数、变量和常量（不能将类放在自定义命名空间中）。

例如，请考虑一个使用 `public` 访问控制命名空间声明的函数。在函数的定义中使用 `public` 属性会将该函数放在 `public` 命名空间中，从而使该函数对于所有的代码都可用。在定义了某个命名空间之后，可以按照与使用 `public` 属性相同的方式来使用所定义的命名空间，该定义将对于可以引用您的自定义命名空间的代码可用。例如，如果您定义一个名为 `example1` 的命名空间，则可以添加一个名为 `myFunction()` 的方法并将 `example1` 用作属性，如下面的示例所示：

```
namespace example1;
class someClass
{
    example1 myFunction() {}
}
```

如果在声明 `myFunction()` 方法时将 `example1` 命名空间用作属性，则意味着该方法属于 `example1` 命名空间。

在应用命名空间时，应切记以下几点：

- 对于每个声明只能应用一个命名空间。
- 不能一次将同一个命名空间属性应用于多个定义。换言之，如果您希望将自己的命名空间应用于 10 个不同的函数，则必须将该命名空间作为属性分别添加到这 10 个函数的定义中。
- 如果您应用了命名空间，则不能同时指定访问控制说明符，因为命名空间和访问控制说明符是互斥的。换言之，如果应用了命名空间，就不能将函数或属性声明为 `public`、`private`、`protected` 或 `internal`。

引用命名空间

在使用借助于任何访问控制命名空间（如 `public`、`private`、`protected` 和 `internal`）声明的方法或属性时，无需显式引用命名空间。这是因为对于这些特殊命名空间的访问由上下文控制。例如，放在 `private` 命名空间中的定义会自动对于同一个类中的代码可用。但是，对于您所定义的命名空间，并不存在这样的上下文相关性。要使用已经放在某个自定义命名空间中的方法或属性，必须引用该命名空间。

可以用 `use namespace` 指令来引用命名空间，也可以使用名称限定符 (`::`) 来以命名空间限定名称。用 `use namespace` 指令引用命名空间会“打开”该命名空间，以便将该命名空间应用于任何未限定的标识符。例如，如果您已经定义了 `example1` 命名空间，则可以通过使用 `use namespace example1` 来访问该命名空间中的名称：

```
use namespace example1;
myFunction();
```

一次可以打开多个命名空间。在使用 `use namespace` 打开了某个命名空间之后，它会在打开它的整个代码块中保持打开状态。不能显式关闭命名空间。

但是，如果同时打开多个命名空间则会增加发生名称冲突的可能性。如果您不愿意打开命名空间，则可以用命名空间和名称限定符来限定方法或属性名，从而避免使用 `use namespace` 指令。例如，下面的代码说明如何用 `example1` 命名空间来限定 `myFunction()` 名称：

```
example1::myFunction();
```

使用命名空间

在包含在 ActionScript 3.0 中的 `flash.utils.Proxy` 类中，可以找到用来防止名称冲突的命名空间的实际示例。`Proxy` 类取代了 ActionScript 2.0 中的 `Object._resolve` 属性，可用来截获对未定义的属性或方法的引用，以免发生错误。为了避免名称冲突，`Proxy` 类的所有方法都位于 `flash_proxy` 命名空间中。

为了更好地了解 `flash_proxy` 命名空间的使用方法，您需要了解如何使用 `Proxy` 类。`Proxy` 类的功能仅对于继承它的类可用。换言之，如果您要对某个对象使用 `Proxy` 类的方法，则该对象的类定义必须扩展 `Proxy` 类。例如，如果您希望截获对未定义的方法的调用，则应扩展 `Proxy` 类，然后覆盖 `Proxy` 类的 `callProperty()` 方法。

前面已讲到，实现命名空间的过程通常分为三步，即定义、应用然后引用命名空间。但是，由于您从不显式调用 `Proxy` 类的任何方法，因此只是定义和应用 `flash_proxy` 命名空间，而从不引用它。ActionScript 3.0 定义 `flash_proxy` 命名空间并在 `Proxy` 类中应用它。在您的代码中，只需要将 `flash_proxy` 命名空间应用于扩展 `Proxy` 类的类。

`flash_proxy` 命名空间按照与下面类似的方法在 `flash.utils` 包中定义：

```
package flash.utils
{
    public namespace flash_proxy;
}
```

该命名空间将应用于 `Proxy` 类的方法，如下面摘自 `Proxy` 类的代码所示：

```
public class Proxy
{
    flash_proxy function callProperty(name:*, ... rest):*
    flash_proxy function deleteProperty(name:*) :Boolean
    ...
}
```

如下面的代码所示，您必须先导入 `Proxy` 类和 `flash_proxy` 命名空间。随后必须声明自己的类，以便它对 `Proxy` 类进行扩展（如果是在严格模式下进行编译，则还必须添加 `dynamic` 属性）。在覆盖 `callProperty()` 方法时，必须使用 `flash_proxy` 命名空间。

```
package
{
    import flash.utils.Proxy;
    import flash.utils.flash_proxy;

    dynamic class MyProxy extends Proxy
    {
        flash_proxy override function callProperty(name:*, ...rest):*
        {
            trace("method call intercepted: " + name);
        }
    }
}
```

如果您创建 `MyProxy` 类的一个实例，并调用一个未定义的方法（如在下面的示例中调用的 `testing()` 方法），则 `Proxy` 对象会截获对该方法的调用，并执行覆盖后的 `callProperty()` 方法内部的语句（在本例中为一个简单的 `trace()` 语句）。

```
var mySample:MyProxy = new MyProxy();
mySample.testing(); // method call intercepted: testing
```

将 `Proxy` 类的方法放在 `flash_proxy` 命名空间内部有两个好处。第一个好处是，在扩展 `Proxy` 类的任何类的公共接口中，拥有单独的命名空间可提高代码的可读性。（在 `Proxy` 类中大约有 12 个可以覆盖的方法，所有这些方法都不能直接调用。将所有这些方法都放在公共命名空间中可能会引起混淆。）第二个好处是，当 `Proxy` 子类中包含名称与 `Proxy` 类方法的名称匹配的实例方法时，使用 `flash_proxy` 命名空间可避免名称冲突。例如，您可能希望将自己的某个方法命名为 `callProperty()`。下面的代码是可接受的，因为您所用的 `callProperty()` 方法位于另一个命名空间中：

```
dynamic class MyProxy extends Proxy
{
    public function callProperty() {}
    flash_proxy override function callProperty(name:*, ...rest):*
    {
        trace("method call intercepted: " + name);
    }
}
```

当您希望以一种无法由四个访问控制说明符（`public`、`private`、`internal` 和 `protected`）实现的方式提供对方法或属性的访问时，命名空间也可能会非常有用。例如，您可能有几个分散在多个包中的实用程序方法。您希望这些方法对于您的所有包均可用，但是不希望这些方法成为公共方法。为此，您可以创建一个新的命名空间，并将它用作您自己的特殊访问控制说明符。

下面的示例使用用户定义的命名空间将两个位于不同包中的函数组合在一起。通过将这两个函数组合到同一个命名空间中，可以通过一条 `use namespace` 语句使这两个函数对于某个类或某个包可见。

本示例使用四个文件来说明此方法。所有的文件都必须位于您的类路径中。第一个文件 (`myInternal.as`) 用来定义 `myInternal` 命名空间。由于该文件位于名为 `example` 的包中，因此您必须将该文件放在名为 `example` 的文件夹中。该命名空间标记为 `public`，因此可以导入到其它包中。

```
// myInternal.as in folder example
package example
{
    public namespace myInternal = "http://www.adobe.com/2006/actionscript/examples";
}
```

第二个文件 (`Utility.as`) 和第三个文件 (`Helper.as`) 定义的类中包含应可供其它包使用的方法。`Utility` 类位于 `example.alpha` 包中，这意味着该文件应放在 `example` 文件夹下的 `alpha` 子文件夹中。`Helper` 类位于 `example.beta` 包中，这意味着该文件应放在 `example` 文件夹下的 `beta` 子文件夹中。这两个包（`example.alpha` 和 `example.beta`）在使用命名空间之前必须先导入它。

```
// Utility.as in the example/alpha folder
package example.alpha
{
    import example.myInternal;

    public class Utility
    {
        private static var _taskCounter:int = 0;

        public static function someTask()
        {
            _taskCounter++;
        }

        myInternal static function get taskCounter():int
        {
            return _taskCounter;
        }
    }
}
```

```
// Helper.as in the example/beta folder
package example.beta
{
    import example.myInternal;

    public class Helper
    {
        private static var _timeStamp:Date;

        public static function someTask()
        {
            _timeStamp = new Date();
        }

        myInternal static function get lastCalled():Date
        {
            return _timeStamp;
        }
    }
}
```

第四个文件 (NamespaceUseCase.as) 是主应用程序类，应是 example 文件夹的同级。在 Adobe Flash CS4 Professional 中，此类用作 FLA 的文档类。NamespaceUseCase 类还导入 myInternal 命名空间，并使用该命名空间调用两个位于其它包中的静态方法。在本示例中，使用静态方法的目的仅在于简化代码。在 myInternal 命名空间中既可以放置静态方法也可以放置实例方法。

```
// NamespaceUseCase.as
package
{
    import flash.display.MovieClip;
    import example.myInternal; // import namespace
    import example.alpha.Utility;// import Utility class
    import example.beta.Helper;// import Helper class

    public class NamespaceUseCase extends MovieClip
    {
        public function NamespaceUseCase()
        {
            use namespace myInternal;

            Utility.someTask();
            Utility.someTask();
            trace(Utility.taskCounter); // 2

            Helper.someTask();
            trace(Helper.lastCalled); // [time someTask() was last called]
        }
    }
}
```

变量

变量可用来存储程序中使用的值。要声明变量，必须将 var 语句和变量名结合使用。在 ActionScript 2.0 中，只有当您使用类型注释时，才需要使用 var 语句。在 ActionScript 3.0 中，总是需要使用 var 语句。例如，下面的 ActionScript 行声明一个名为 i 的变量：

```
var i;
```

如果在声明变量时省略了 var 语句，则在严格模式下会出现编译器错误，在标准模式下会出现运行时错误。例如，如果以前未定义变量 i，则下面的代码行将产生错误：

```
i; // error if i was not previously defined
```

要将变量与一个数据类型相关联，则必须在声明变量时进行此操作。在声明变量时不指定变量的类型是合法的，但这在严格模式下会产生编译器警告。可通过在变量名后面追加一个后跟变量类型的冒号 (:) 来指定变量类型。例如，下面的代码声明一个 int 类型的变量 i：

```
var i:int;
```

可以使用赋值运算符 (=) 为变量赋值。例如，下面的代码声明一个变量 i 并将值 20 赋给它：

```
var i:int;
i = 20;
```

您可能会发现在声明变量的同时为变量赋值可能更加方便，如下面的示例所示：

```
var i:int = 20;
```

通常，在声明变量的同时为变量赋值的方法不仅在赋予基元值（如整数和字符串）时很常用，而且在创建数组或实例化类的实例时也很常用。下面的示例显示了一个使用一行代码声明和赋值的数组。

```
var numArray:Array = ["zero", "one", "two"];
```

可以使用 new 运算符来创建类的实例。下面的示例创建一个名为 CustomClass 的实例，并向名为 customItem 的变量赋予对该实例的引用：

```
var customItem:CustomClass = new CustomClass();
```

如果要声明多个变量，则可以使用逗号运算符 (,) 来分隔变量，从而在一行代码中声明所有这些变量。例如，下面的代码在一行代码中声明 3 个变量：

```
var a:int, b:int, c:int;
```

也可以在同一行代码中为其中的每个变量赋值。例如，下面的代码声明 3 个变量（a、b 和 c）并为每个变量赋值：

```
var a:int = 10, b:int = 20, c:int = 30;
```

尽管您可以使用逗号运算符来将各个变量的声明组合到一条语句中，但是这样可能会降低代码的可读性。

了解变量的作用域

变量的范围是指可在其中通过词汇引用来自访问变量的代码区域。全局变量是指在代码的所有区域中定义的变量，而局部变量是指仅在代码的某个部分定义的变量。在 ActionScript 3.0 中，始终为变量分配声明它们的函数或类的作用域。全局变量是在任何函数或类定义的外部定义的变量。例如，下面的代码通过在任何函数的外部声明一个名为 strGlobal 的全局变量来创建该变量。从该示例可看出，全局变量在函数定义的内部和外部均可用。

```
var strGlobal:String = "Global";
function scopeTest()
{
    trace(strGlobal); // Global
}
scopeTest();
trace(strGlobal); // Global
```

可以通过在函数定义内部声明变量来将它声明为局部变量。可定义局部变量的最小代码区域就是函数定义。在函数内部声明的局部变量仅存在于该函数中。例如，如果在名为 localScope() 的函数中声明一个名为 str2 的变量，该变量在该函数外部将不可用。

```

function localScope()
{
    var strLocal:String = "local";
}
localScope();
trace(strLocal); // error because strLocal is not defined globally

```

如果用于局部变量的变量名已经被声明为全局变量，那么，当局部变量在作用域内时，局部定义会隐藏（或遮蔽）全局定义。全局变量在该函数外部仍然存在。例如，下面的代码创建一个名为 str1 的全局字符串变量，然后在 scopeTest() 函数内部创建一个同名的局部变量。该函数中的 trace 语句输出该变量的局部值，而函数外部的 trace 语句则输出该变量的全局值。

```

var str1:String = "Global";
function scopeTest ()
{
    var str1:String = "Local";
    trace(str1); // Local
}
scopeTest();
trace(str1); // Global

```

与 C++ 和 Java 中的变量不同的是，ActionScript 变量没有块级作用域。代码块是指左大括号 ({}) 与右大括号 (}) 之间的任意一组语句。在某些编程语言（如 C++ 和 Java）中，在代码块内部声明的变量在代码块外部不可用。对于作用域的这一限制称为块级作用域，ActionScript 中不存在这样的限制，如果您在某个代码块中声明一个变量，那么，该变量不仅在该代码块中可用，而且还在该代码块所属函数的其它任何部分都可用。例如，下面的函数包含在不同的块作用域中定义的变量。所有的变量均在整个函数中可用。

```

function blockTest (testArray:Array)
{
    var numElements:int = testArray.length;
    if (numElements > 0)
    {
        var elemStr:String = "Element #";
        for (var i:int = 0; i < numElements; i++)
        {
            var valueStr:String = i + ": " + testArray[i];
            trace(elemStr + valueStr);
        }
        trace(elemStr, valueStr, i); // all still defined
    }
    trace(elemStr, valueStr, i); // all defined if numElements > 0
}

blockTest(["Earth", "Moon", "Sun"]);

```

有趣的是，如果缺乏块级作用域，那么，只要在函数结束之前对变量进行声明，就可以在声明变量之前读写它。这是因为存在一种名为提升的方法，该方法表示编译器会将所有的变量声明移到函数的顶部。例如，下面的代码会进行编译，即使先有 num 变量的初始 trace() 函数后声明 num 变量也是如此：

```

trace(num); // NaN
var num:Number = 10;
trace(num); // 10

```

但是，编译器将不会提升任何赋值语句。这就说明了为什么 num 的初始 trace() 会生成 NaN（而非某个数字），NaN 是 Number 数据类型变量的默认值。这意味着您甚至可以在声明变量之前为变量赋值，如下面的示例所示：

```

num = 5;
trace(num); // 5
var num:Number = 10;
trace(num); // 10

```

默认值

默认值是设置变量的值之前变量中包含的值。首次设置变量的值实际上就是初始化变量。如果您声明了一个变量，但是没有设置它的值，则该变量便处于未初始化状态。未初始化的变量的值取决于它的数据类型。下表说明了变量的默认值，并按数据类型对这些值进行组织：

数据类型	默认值
Boolean	false
int	0
Number	NaN
Object	null
String	null
uint	0
未声明（与类型注释 * 等效）	undefined
其它所有类（包括用户定义的类）。	null

对于 Number 类型的变量，默认值是 NaN（而非某个数字），NaN 是一个由 IEEE-754 标准定义的特殊值，它表示非数字的某个值。

如果您声明某个变量，但是未声明它的数据类型，则将应用默认数据类型 *，这实际上表示该变量是无类型变量。如果您还没有用值初始化无类型变量，则该变量的默认值是 undefined。

对于 Boolean、Number、int 和 uint 以外的数据类型，所有未初始化变量的默认值都是 null。这适用于由 ActionScript 3.0 定义的所有类以及您创建的所有自定义类。

对于 Boolean、Number、int 或 uint 类型的变量，null 不是有效值。如果您尝试将值 null 赋予这样的变量，则该值会转换为该数据类型的默认值。对于 Object 类型的变量，可以赋予 null 值。如果您尝试将值 undefined 赋予 Object 类型的变量，则该值会转换为 null。

对于 Number 类型的变量，有一个名为 isNaN() 的特殊的顶级函数，如果变量不是数字，则该函数会返回布尔值 true，否则会返回 false。

数据类型

数据类型用于定义一组值。例如，Boolean 数据类型所定义的一组值中仅包含两个值：true 和 false。除了 Boolean 数据类型外，ActionScript 3.0 还定义了其它几个常用的数据类型，如 String、Number 和 Array。您可以使用类或接口来自定义一组值，从而定义自己的数据类型。ActionScript 3.0 中的所有值均是对象，无论这些值是基元值还是复杂值。

基元值是一个属于下列数据类型之一的值：Boolean、int、Number、String 和 uint。基元值的处理速度通常比复杂值的处理速度快，因为 ActionScript 按照一种尽可能优化内存和提高速度的特殊方式来存储基元值。

注：关注技术细节的读者会发现，ActionScript 在内部将基元值作为不可改变的对象进行存储。这意味着按引用传递与按值传递同样有效。这可以减少内存的使用量并提高执行速度，因为引用通常比值本身小得多。

复杂值是指基元值以外的值。定义复杂值的集合的数据类型包括：Array、Date、Error、Function、RegExp、XML 和 XMLList。

许多编程语言都区分基元值及其包装对象。例如，Java 中有一个 int 基元值和一个包装它的 java.lang.Integer 类。Java 基元值不是对象，但它们的包装是对象，这使得基元值对于某些运算非常有用，而包装对象则更适合于其它运算。在 ActionScript 3.0 中，出于实用的目的，不对基元值及其包装对象加以区分。所有的值（甚至基元值）都是对象。Flash Player 和 Adobe AIR 将这些基元类型视为特例，它们的行为与对象相似，但是不需要创建对象所涉及的正常开销。这意味着下面的两行代码是等效的：

```
var someInt:int = 3;
var someInt:int = new int(3);
```

上面列出的所有基元数据类型和复杂数据类型都是由 ActionScript 3.0 核心类定义的。通过 ActionScript 3.0 核心类，可以使用字面值（而非 new 运算符）创建对象。例如，可以使用字面值或 Array 类的构造函数来创建数组，如下所示：

```
var someArray:Array = [1, 2, 3]; // literal value
var someArray:Array = new Array(1,2,3); // Array constructor
```

类型检查

类型检查可以在编译时或运行时执行。静态类型语言（如 C++ 和 Java）在编译时执行类型检查。动态类型语言（如 Smalltalk 和 Python）在运行时执行类型检查。作为动态类型的语言，ActionScript 3.0 在运行时执行类型检查，同时也支持在称为严格模式的特殊编译器模式下在编译时执行类型检查。在严格模式下，类型检查既发生在编译时也发生在运行时，但是在标准模式下，类型检查仅发生在运行时。

在构造代码时，动态类型的语言带来了极大的灵活性，但代价是在运行时可能出现类型错误。静态类型的语言在编译时报告类型错误，但代价是要求类型信息在编译时是已知的。

编译时类型检查

在较大的项目中通常建议使用编译时类型检查，因为随着项目变大，相对于尽早捕获类型错误，数据类型的灵活性通常会变得不那么重要。这就是为什么将 Adobe Flash CS4 Professional 和 Adobe Flex Builder 中的 ActionScript 编译器默认设置为在严格模式下运行的原因。

为了提供编译时类型检查，编译器需要知道代码中的变量或表达式的数据类型信息。为了显式声明变量的数据类型，请在变量名后面添加后跟数据类型的冒号运算符（:）作为其后缀。要将数据类型与参数相关联，应使用后跟数据类型的冒号运算符。例如，下面的代码向 xParam 参数中添加数据类型信息，并用显式数据类型声明变量 myParam：

```
function runtimeTest(xParam:String)
{
    trace(xParam);
}
var myParam:String = "hello";
runtimeTest(myParam);
```

在严格模式下，ActionScript 编译器将类型不匹配报告为编译器错误。例如，下面的代码声明一个 Object 类型的函数参数 xParam，但是之后又尝试向该参数赋予 String 类型和 Number 类型的值。这在严格模式下会产生编译器错误。

```
function dynamicTest(xParam:Object)
{
    if (xParam is String)
    {
        var myStr:String = xParam; // compiler error in strict mode
        trace("String: " + myStr);
    }
    else if (xParam is Number)
    {
        var myNum:Number = xParam; // compiler error in strict mode
        trace("Number: " + myNum);
    }
}
```

但是，即使在严格模式下，也可以选择不在赋值语句右侧指定类型，从而退出编译时类型检查。可以通过省略类型注释或使用特殊的星号(*)类型注释，来将变量或表达式标记为无类型。例如，如果对上一个示例中的 xParam 参数进行修改，以便其不再有类型注释，则代码会在严格模式下进行编译：

```
function dynamicTest(xParam)
{
    if (xParam is String)
    {
        var myStr:String = xParam;
        trace("String: " + myStr);
    }
    else if (xParam is Number)
    {
        var myNum:Number = xParam;
        trace("Number: " + myNum);
    }
}
dynamicTest(100);
dynamicTest("one hundred");
```

运行时类型检查

在 ActionScript 3.0 中，无论是在严格模式下还是在标准模式下编译，在运行时都将进行类型检查。请考虑以下情形：将值 3 作为一个参数传递给需要数组的函数。在严格模式下，编译器将生成一个错误，因为值 3 与 Array 数据类型不兼容。如果您禁用了严格模式而在标准模式下运行，则编译器不会指出类型不匹配，但是当 Flash Player 和 Adobe AIR 在运行时进行类型检查时则会产生运行时错误。

下面的示例演示一个名为 typeTest() 的函数，该函数需要一个 Array 参数，但传递的值却是 3。在标准模式下这会产生运行时错误，因为值 3 不是参数声明的数据类型 (Array) 的成员。

```
function typeTest(xParam:Array)
{
    trace(xParam);
}
var myNum:Number = 3;
typeTest(myNum);
// run-time error in ActionScript 3.0 standard mode
```

还可能会出现如下情形：即使在严格模式下运行，也可能会获得运行时类型错误。如果您使用严格模式，但是通过使用无类型变量而退出了编译时类型检查，就可能会出现上述情形。当您使用无类型变量时，并不会消除类型检查，而只是将其延迟到运行时执行。例如，如果上一个示例中的 myNum 变量没有已声明的数据类型，则编译器会无法检测到类型不匹配，但是 Flash Player 和 Adobe AIR 会生成一个运行时错误，因为其会将 myNum 的运行时值（赋值语句将其设置为 3）与 xParam 的类型（设置为 Array 数据类型）进行比较。

```
function typeTest(xParam:Array)
{
    trace(xParam);
}
var myNum = 3;
typeTest(myNum);
// run-time error in ActionScript 3.0
```

与编译时类型检查相比，运行时类型检查还允许更灵活地使用继承。标准模式会将类型检查延迟到运行时执行，从而使您可以引用子类的属性，即使您上传也是如此。当您使用基类来声明类实例的类型，但是使用子类来实例化类实例时，就会发生上传。例如，您可以创建一个名为 ClassBase 的可扩展类（具有 final 属性的类不能扩展）：

```
class ClassBase
{
}
```

随后，您可以创建一个名为 ClassExtender 的 ClassBase 子类，该子类具有一个名为 someString 的属性，如下所示：

```
class ClassExtender extends ClassBase
{
    var someString:String;
}
```

通过使用这两个类，可以创建一个使用 `ClassBase` 数据类型进行声明、但使用 `ClassExtender` 构造函数进行实例化的类实例。上传被视为安全操作，这是因为基类不包含子类中没有的任何属性或方法。

```
var myClass:ClassBase = new ClassExtender();
```

但是，子类中则包含其基类中没有的属性或方法。例如，`ClassExtender` 类中包含 `someString` 属性，该属性在 `ClassBase` 类中不存在。在 ActionScript 3.0 标准模式下，可以使用 `myClass` 实例来引用此属性，而不会生成编译时错误，如下面的示例所示：

```
var myClass:ClassBase = new ClassExtender();
myClass.someString = "hello";
// no error in ActionScript 3.0 standard mode
```

is 运算符

`is` 运算符是 ActionScript 3.0 中的新增运算符，它可用来测试变量或表达式是否为给定数据类型的成员。在早期的 ActionScript 版本中，此功能由 `instanceof` 运算符提供。但在 ActionScript 3.0 中，不应使用 `instanceof` 运算符来测试变量或表达式是否为数据类型的成员。对于手动类型检查，应用 `is` 运算符来代替 `instanceof` 运算符，因为表达式 `x instanceof y` 只是在 `x` 的原型链中检查 `y` 是否存在（在 ActionScript 3.0 中，原型链不能全面地描述继承层次结构）。

`is` 运算符检查正确的继承层次结构，不但可以用来检查对象是否为特定类的实例，而且还可以检查对象是否是用来实现特定接口的类的实例。下面的示例创建 `Sprite` 类的一个名为 `mySprite` 的实例，并使用 `is` 运算符来测试 `mySprite` 是否为 `Sprite` 和 `DisplayObject` 类的实例，以及它是否实现 `IEventDispatcher` 接口：

```
var mySprite:Sprite = new Sprite();
trace(mySprite is Sprite); // true
trace(mySprite is DisplayObject); // true
trace(mySprite is IEventDispatcher); // true
```

`is` 运算符检查继承层次结构，并正确地报告 `mySprite` 与 `Sprite` 和 `DisplayObject` 类兼容（`Sprite` 类是 `DisplayObject` 类的子类）。`is` 运算符还检查 `mySprite` 是否是从实现 `IEventDispatcher` 接口的任意类继承的。由于 `Sprite` 类是从实现 `IEventDispatcher` 接口的 `EventDispatcher` 类继承的，因此 `is` 运算符会正确地报告 `mySprite` 也实现该接口。

下面的示例说明与上一个示例相同的测试，但使用的是 `instanceof` 运算符，而不是 `is` 运算符。`instanceof` 运算符正确地识别出 `mySprite` 是 `Sprite` 或 `DisplayObject` 的实例，但是，当该运算符用于测试 `mySprite` 是否实现 `IEventDispatcher` 接口时，返回的却是 `false`。

```
trace(mySprite instanceof Sprite); // true
trace(mySprite instanceof DisplayObject); // true
trace(mySprite instanceof IEventDispatcher); // false
```

as 运算符

`as` 运算符是 ActionScript 3.0 中的新增运算符，也可用来检查表达式是否为给定数据类型的成员。但是，与 `is` 运算符不同的是，`as` 运算符不返回布尔值，`as` 运算符返回表达式的值（代替 `true`）或 `null`（代替 `false`）。下面的示例演示在检查 `Sprite` 实例是否为 `DisplayObject`、`IEventDispatcher` 和 `Number` 数据类型的成员这样的简单情况下，使用 `as` 运算符替代 `is` 运算符的结果。

```
var mySprite:Sprite = new Sprite();
trace(mySprite as Sprite); // [object Sprite]
trace(mySprite as DisplayObject); // [object Sprite]
trace(mySprite as IEventDispatcher); // [object Sprite]
trace(mySprite as Number); // null
```

在使用 `as` 运算符时，右侧的操作数必须是数据类型。如果尝试使用表达式（而非数据类型）作为右侧的操作数，将会产生错误。

动态类

动态类定义在运行时可通过添加或更改属性和方法来改变的对象。非动态类（如 String 类）是密封类。您不能在运行时向密封类中添加属性或方法。

在声明类时，可以通过使用 dynamic 属性来创建动态类。例如，下面的代码创建一个名为 Protean 的动态类：

```
dynamic class Protean
{
    private var privateGreeting:String = "hi";
    public var publicGreeting:String = "hello";
    function Protean()
    {
        trace("Protean instance created");
    }
}
```

如果要在以后实例化 Protean 类的实例，则可以在类定义的外部向其添加属性或方法。例如，下面的代码创建 Protean 类的一个实例，并向该实例中添加两个名称分别为 aString 和 aNumber 的属性：

```
var myProtean:Protean = new Protean();
myProtean.aString = "testing";
myProtean.aNumber = 3;
trace(myProtean.aString, myProtean.aNumber); // testing 3
```

添加到动态类实例中的属性是运行时实体，因此会在运行时完成所有类型检查。不能向以这种方式添加的属性中添加类型注释。

您还可以定义一个函数并将该函数附加到 myProtean 实例的某个属性，从而向 myProtean 实例中添加方法。下面的代码将 trace 语句移到一个名为 traceProtean() 的方法中：

```
var myProtean:Protean = new Protean();
myProtean.aString = "testing";
myProtean.aNumber = 3;
myProtean.traceProtean = function ()
{
    trace(this.aString, this.aNumber);
};
myProtean.traceProtean(); // testing 3
```

但是，以这种方式创建的方法对于 Protean 类的任何私有属性或方法都不具有访问权限。而且，即使对 Protean 类的公共属性或方法的引用也必须用 this 关键字或类名进行限定。下面的示例说明了 traceProtean() 方法，该方法尝试访问 Protean 类的私有变量和公共变量。

```
myProtean.traceProtean = function ()
{
    trace(myProtean.privateGreeting); // undefined
    trace(myProtean.publicGreeting); // hello
};
myProtean.traceProtean();
```

数据类型说明

基元数据类型包括 Boolean、int、Null、Number、String、uint 和 void。ActionScript 核心类还定义下列复杂数据类型：Object、Array、Date、Error、Function、RegExp、XML 和 XMLList。

Boolean 数据类型

Boolean 数据类型包含两个值：true 和 false。对于 Boolean 类型的变量，其它任何值都是无效的。已经声明但尚未初始化的布尔变量的默认值是 false。

int 数据类型

int 数据类型在内部存储为 32 位整数，包含的整数集介于

-2,147,483,648 (- 2^{31}) 至 2,147,483,647 ($2^{31} - 1$) 之间（两端包含在内）。早期的 ActionScript 版本仅提供 Number 数据类型，该数据类型既可用于整数又可用于浮点数。在 ActionScript 3.0 中，现在可以访问 32 位带符号整数和无符号整数的低位机器类型。如果您的变量将不会使用浮点数，那么，使用 int 数据类型来代替 Number 数据类型应会更快更高效。

对于小于 int 的最小值或大于 int 的最大值的整数值，应使用 Number 数据类型。Number 数据类型可以处理 -9,007,199,254,740,992 和 9,007,199,254,740,992 (53 位整数值) 之间的值。int 数据类型的变量的默认值是 0。

Null 数据类型

Null 数据类型仅包含一个值：null。这是 String 数据类型和用来定义复杂数据类型的所有类（包括 Object 类）的默认值。其它基元数据类型（如 Boolean、Number、int 和 uint）均不包含 null 值。如果您尝试向 Boolean、Number、int 或 uint 类型的变量赋予 null，则 Flash Player 和 Adobe AIR 会将 null 值转换为相应的默认值。不能将 Null 数据类型用作类型注释。

Number 数据类型

在 ActionScript 3.0 中，Number 数据类型可以表示整数、无符号整数和浮点数。但是，为了尽可能提高性能，应将 Number 数据类型仅用于浮点数，或者用于 int 和 uint 类型可以存储的、大于 32 位的整数值。要存储浮点数，数字中应包括一个小数点。如果您省略了小数点，数字将存储为整数。

Number 数据类型使用由 IEEE 二进制浮点算术标准 (IEEE-754) 指定的 64 位双精度格式。此标准规定如何使用 64 个可用位来存储浮点数。其中的 1 位用来指定数字是正数还是负数。11 位用于指数，以 2 为底进行存储。其余的 52 位用于存储有效位数（又称为尾数），有效位数是 2 的 N 次幂，N 即前面所提到的指数。

可以将 Number 数据类型的所有位都用于有效位数，也可以将 Number 数据类型的某些位用于存储指数，后者可存储的浮点数比前者大得多。例如，如果 Number 数据类型使用全部 64 位存储有效位数，则可以存储的最大数字为 $2^{65} - 1$ 。如果使用 11 位存储指数，则 Number 数据类型可以存储的最大有效数字为 2^{1023} 。

Number 类型可以表示的最大值和最小值存储在 Number 类的名为 Number.MAX_VALUE 和 Number.MIN_VALUE 的静态属性中。

```
Number.MAX_VALUE == 1.79769313486231e+308
Number.MIN_VALUE == 4.940656458412467e-324
```

尽管这个数字范围很大，但代价是此范围的精度有所降低。Number 数据类型使用 52 位来存储有效位数，因此，那些要求用 52 位以上的位数才能精确表示的数字（如分数 1/3）将只是近似值。如果应用程序要求小数达到绝对精度，则需要使用实现小数浮点算术（而非二进制浮点算术）的软件。

如果用 Number 数据类型来存储整数值，则仅使用 52 位有效位数。Number 数据类型使用这 52 位和一个特殊的隐藏位来表示从 -9,007,199,254,740,992 (- 2^{53}) 到 9,007,199,254,740,992 (2^{53}) 的整数。

Flash Player 和 Adobe AIR 不但将 NaN 值用作 Number 类型的变量的默认值，而且还将其用作应返回数字、却没有返回数字的任何运算的结果。例如，如果您尝试计算负数的平方根，结果将是 NaN。其它特殊的 Number 值包括正无穷大 和负无穷大。

注：在被 0 除时，如果被除数也是 0，则结果只有一个，那就是 NaN。在被 0 除时，如果被除数是正数，则结果为“正无穷大”；如果被除数是负数，则结果为“负无穷大”。

String 数据类型

String 数据类型表示一个 16 位字符的序列。字符串在内部存储为 Unicode 字符，并使用 UTF-16 格式。字符串是不可改变的值，就像在 Java 编程语言中一样。对字符串值的操作返回字符串的一个新的实例。用字符串数据类型声明的变量的默认值是 null。虽然 null 值与空字符串 ("") 均表示没有任何字符，但二者并不相同。

uint 数据类型

`uint` 数据类型在内部存储为 32 位无符号整数，包含的整数集介于 0 和 4,294,967,295 ($2^{32} - 1$) 之间（包括 0 和 4,294,967,295）。`uint` 数据类型可用于要求非负整数的特殊情形。例如，必须使用 `uint` 数据类型来表示像素颜色值，因为 `int` 数据类型有一个内部符号位，该符号位并不适合处理颜色值。对于大于 `uint` 的最大值的整数值，应使用 `Number` 数据类型，该数据类型可以处理 53 位整数值。`uint` 数据类型的变量的默认值是 0。

void 数据类型

`void` 数据类型仅包含一个值：`undefined`。在早期的 ActionScript 版本中，`undefined` 是 `Object` 类实例的默认值。在 ActionScript 3.0 中，`Object` 实例的默认值是 `null`。如果您尝试将值 `undefined` 赋予 `Object` 类的实例，则 Flash Player 或 Adobe AIR 会将该值转换为 `null`。您只能为无类型变量赋予 `undefined` 这一值。无类型变量是指缺乏类型注释或者使用星号 (*) 作为类型注释的变量。您可以将 `void` 只用作返回类型注释。

Object 数据类型

`Object` 数据类型是由 `Object` 类定义的。`Object` 类用作 ActionScript 中的所有类定义的基类。ActionScript 3.0 中的 `Object` 数据类型与早期版本中的 `Object` 数据类型存在以下三方面的区别：第一，`Object` 数据类型不再是指定给没有类型注释的变量的默认数据类型。第二，`Object` 数据类型不再包括 `undefined` 这一值，该值以前是 `Object` 实例的默认值。第三，在 ActionScript 3.0 中，`Object` 类实例的默认值是 `null`。

在早期的 ActionScript 版本中，会自动为没有类型注释的变量赋予 `Object` 数据类型。ActionScript 3.0 现在包括真正无类型变量这一概念，因此不再为没有类型注释的变量赋予 `Object` 数据类型。没有类型注释的变量现在被视为无类型变量。如果您希望向代码的读者清楚地表明您是故意将变量保留为无类型，可以使用新的星号 (*) 表示类型注释，这与省略类型注释等效。下面的示例演示两条等效的语句，两者都声明一个无类型变量 `x`：

```
var x
var x:*
```

只有无类型变量才能保存值 `undefined`。如果您尝试将值 `undefined` 赋给具有数据类型的变量，则 Flash Player 或 Adobe AIR 会将值 `undefined` 转换为该数据类型的默认值。对于 `Object` 数据类型的实例，默认值是 `null`，这意味着，如果尝试将 `undefined` 赋给 `Object` 实例，则 Flash Player 或 Adobe AIR 会将值 `undefined` 转换为 `null`。

类型转换

在将某个值转换为其它数据类型的值时，就说发生了类型转换。类型转换可以是隐式的，也可以是显式的。隐式转换又称为强制，有时由 Flash Player 或 Adobe AIR 在运行时执行。例如，如果将值 2 赋给 `Boolean` 数据类型的变量，则 Flash Player 或 Adobe AIR 会先将值 2 转换为布尔值 `true`，然后再将其赋给该变量。显式转换又称为转换，在代码指示编译器将一个数据类型的变量视为属于另一个数据类型时发生。在涉及基元值时，转换功能将一个数据类型的值实际转换为另一个数据类型的值。要将对象转换为另一类型，请用小括号括起对象名并在它前面加上新类型的名称。例如，下面的代码提取一个布尔值并将它转换为一个整数：

```
var myBoolean:Boolean = true;
var myINT:int = int(myBoolean);
trace(myINT); // 1
```

隐式转换

在运行时，隐式转换会发生在许多上下文中：

- 在赋值语句中
- 在将值作为函数的参数传递时
- 在从函数中返回值时
- 在使用某些运算符（如加法运算符 (+)）的表达式中

对于用户定义的类型，当要转换的值是目标类（或者派生自目标类的类）的实例时，隐式转换会成功。如果隐式转换不成功，就会出现错误。例如，下面的代码中包含成功的隐式转换和不成功的隐式转换：

```
class A {}
class B extends A {}

var objA:A = new A();
var objB:B = new B();
var arr:Array = new Array();

objA = objB; // Conversion succeeds.
objB = arr; // Conversion fails.
```

对于基元类型而言，隐式转换是通过调用内部转换算法来处理的，该算法与显式转换函数所调用的算法相同。下面各部分详细讨论了这些基元类型转换。

显式转换

在严格模式下进行编译时，使用显式转换会非常有用，因为您有时可能不希望因类型不匹配而生成编译时错误。当您知道强制功能会在运行时正确转换您的值时，可能就属于这种情况。例如，在处理从表单接收的数据时，您可能希望依赖强制功能将某些字符串值转换为数值。下面的代码会生成编译时错误，即使代码在标准模式下能够正确运行也是如此：

```
var quantityField:String = "3";
var quantity:int = quantityField; // compile time error in strict mode
```

如果您希望继续使用严格模式，但是希望将字符串转换为整数，则可以使用显式转换，如下所示：

```
var quantityField:String = "3";
var quantity:int = int(quantityField); // Explicit conversion succeeds.
```

转换为 int、uint 和 Number

您可以将任何数据类型转换为以下三种数字类型之一：int、uint 和 Number。如果 Flash Player 或 Adobe AIR 由于某种原因无法转换数字，则会为 int 和 uint 数据类型赋予默认值 0，为 Number 数据类型赋予默认值 NaN。如果将布尔值转换为数字，则 true 变成值 1，false 变成值 0。

```
var myBoolean:Boolean = true;
var myUINT:uint = uint(myBoolean);
var myINT:int = int(myBoolean);
var myNum:Number = Number(myBoolean);
trace(myUINT, myINT, myNum); // 1 1 1
myBoolean = false;
myUINT = uint(myBoolean);
myINT = int(myBoolean);
myNum = Number(myBoolean);
trace(myUINT, myINT, myNum); // 0 0 0
```

仅包含数字的字符串值可以成功地转换为数字类型之一。看上去像负数的字符串或者表示十六进制值的字符串（例如，0x1A）也可以转换为数字类型。转换过程中会忽略字符串值中的前导或尾随空白字符。还可以使用 Number() 来转换看上去像浮点数的字符串。如果包含小数点，则会导致 uint() 和 int() 返回一个整数，小数点和它后面的字符被截断。例如，下面的字符串值可以转换为数字：

```
trace(uint("5")); // 5
trace(uint("-5")); // 4294967291. It wraps around from MAX_VALUE
trace(uint(" 27 ")); // 27
trace(uint("3.7")); // 3
trace(int("3.7")); // 3
trace(int("0x1A")); // 26
trace(Number("3.7")); // 3.7
```

对于包含非数字字符的字符串值，在用 int() 或 uint() 转换时，将返回 0；在用 Number() 转换时，将返回 NaN。转换过程中会忽略前导和尾随空白，但是，如果字符串中包含将两个数字隔开的空白，则将返回 0 或 NaN。

```
trace(uint("5a")); // 0
trace(uint("ten")); // 0
trace(uint("17 63")); // 0
```

在 ActionScript 3.0 中，Number() 函数不再支持八进制数或基数为 8 的数。如果您向 ActionScript 2.0 Number() 函数提供的字符串中包含前导 0，则该数字将被视为八进制数，并将转换为等效的十进制数。对于 ActionScript 3.0 中的 Number() 函数，情况并非如此，因为 ActionScript 3.0 会忽略前导 0。例如，在使用不同的 ActionScript 版本进行编译时，下面的代码会生成不同的输出结果：

```
trace(Number("044"));
// ActionScript 3.0 44
// ActionScript 2.0 36
```

将一种数值类型的值赋给另一种数值类型的变量时，转换并不是必需的。即使在严格模式下，数值类型也会隐式转换为其它数值类型。这意味着，在某些情况下，在超出类型的范围时，可能会生成意外的值。下面的几个示例全部是在严格模式下进行编译的，但是某些示例将生成意外的值：

```
var myUInt:uint = -3; // Assign int/Number value to uint variable
trace(myUInt); // 4294967293

var myNum:Number = sampleUINT; // Assign int/uint value to Number variable
trace(myNum) // 4294967293

var myInt:int = uint.MAX_VALUE + 1; // Assign Number value to uint variable
trace(myInt); // 0

myInt = int.MAX_VALUE + 1; // Assign uint/Number value to int variable
trace(myInt); // -2147483648
```

下表概述了将其它数据类型转换为 Number、int 或 uint 数据类型的结果。

数据类型或值	转换为 Number、int 或 uint 时的结果
Boolean	如果值为 true，则结果为 1；否则为 0。
Date	Date 对象的内部表示形式，即从 1970 年 1 月 1 日午夜（通用时间）以来所经过的毫秒数。
null	0
Object	如果实例为 null 并转换为 Number，则结果为 NaN；否则为 0。
String	如果 Flash Player 或 Adobe AIR 可以将字符串转换为数字，则结果为数字；否则，如果转换为 Number，则结果为 NaN，如果转换为 int 或 uint，则结果为 0。
undefined	如果转换为 Number，则结果为 NaN；如果转换为 int 或 uint，则结果为 0。

转换为 Boolean

在从任何数值数据类型（uint、int 和 Number）转换为 Boolean 时，如果数值为 0，则结果为 false；否则为 true。对于 Number 数据类型，如果值为 NaN，则结果也为 false。下面的示例说明在转换 -1、0 和 1 等数字时的结果：

```
var myNum:Number;
for (myNum = -1; myNum<2; myNum++)
{
    trace("Boolean(" + myNum +") is " + Boolean(myNum));
}
```

本示例的输出结果演示在这三个数字中，只有 0 返回 false 值：

```
Boolean(-1) is true
Boolean(0) is false
Boolean(1) is true
```

在将字符串值转换为 Boolean 数据类型时，如果字符串为 null 或空字符串（""），则会返回 false。否则，将返回 true。

```

var str1:String; // Uninitialized string is null.
trace(Boolean(str1)); // false

var str2:String = ""; // empty string
trace(Boolean(str2)); // false

var str3:String = " "; // white space only
trace(Boolean(str3)); // true

```

在将 **Object** 类的实例转换为 **Boolean** 数据类型时，如果该实例为 **null**，则将返回 **false**；否则将返回 **true**：

```

var myObj:Object; // Uninitialized object is null.
trace(Boolean(myObj)); // false

myObj = new Object(); // instantiate
trace(Boolean(myObj)); // true

```

在严格模式下，系统会对布尔变量进行特殊处理，因为您不必转换即可向布尔变量赋予任何数据类型的值。即使在严格模式下，也可以将所有的数据类型隐式强制为 **Boolean** 数据类型。换言之，与几乎其它所有数据类型不同，转换为 **Boolean** 数据类型不是避免在严格模式下出错所必需的。下面的几个示例全部是在严格模式下编译的，它们在运行时按照预期的方式工作：

```

var myObj:Object = new Object(); // instantiate
var bool:Boolean = myObj;
trace(bool); // true
bool = "random string";
trace(bool); // true
bool = new Array();
trace(bool); // true
bool = NaN;
trace(bool); // false

```

下表概述了在从其它数据类型转换为 **Boolean** 数据类型时的结果：

数据类型或值	转换为 Boolean 数据类型时的结果
String	如果值为 null 或空字符串 ("")，则结果为 false ；否则为 true 。
null	false
Number 、 int 或 uint	如果值为 Nan 或 0 ，则结果为 false ；否则为 true 。
Object	如果实例为 null ，则结果为 false ；否则为 true 。

转换为 **String**

从任何数值数据类型转换为 **String** 数据类型时，都会返回数字的字符串表示形式。在将布尔值转换为 **String** 数据类型时，如果值为 **true**，则返回字符串 "true"；如果值为 **false**，则返回字符串 "false"。

在从 **Object** 类的实例转换为 **String** 数据类型时，如果该实例为 **null**，则返回字符串 "null"。否则，从 **Object** 类转换为 **String** 类型会返回字符串 "[object Object]"。

在从 **Array** 类的实例转换为 **String** 时，会返回一个字符串，其中包含所有数组元素的逗号分隔列表。例如，在下面的示例中，在转换为 **String** 数据类型时，将返回一个包含数组中的全部三个元素的字符串：

```

var myArray:Array = ["primary", "secondary", "tertiary"];
trace(String(myArray)); // primary,secondary,tertiary

```

在从 **Date** 类的实例转换为 **String** 时，会返回该实例所包含日期的字符串表示形式。例如，下面的示例返回 **Date** 类实例的字符串表示形式（输出结果显示的是太平洋夏令时）：

```

var myDate:Date = new Date(2005,6,1);
trace(String(myDate)); // Fri Jul 1 00:00:00 GMT-0700 2005

```

下表概述了在将其它数据类型转换为 **String** 数据类型时的结果：

数据类型或值	转换为 String 数据类型时的结果
Array	一个包含所有数组元素的字符串。
Boolean	"true" 或 "false"
Date	Date 对象的字符串表示形式。
null	"null"
Number、int 或 uint	数字的字符串表示形式。
Object	如果实例为 null，则结果为 "null"；否则为 "[object Object]"。

语法

语言的语法定义了一组在编写可执行代码时必须遵循的规则。

区分大小写

ActionScript 3.0 是一种区分大小写的语言。只是大小写不同的标识符会被视为不同。例如，下面的代码创建两个不同的变量：

```
var num1:int;
var Num1:int;
```

点语法

点运算符(.) 提供对对象的属性和方法的访问。使用点语法，可以使用后跟点运算符和属性名或方法名的实例名来引用类的属性或方法。以下面的类定义为例：

```
class DotExample
{
    public var prop1:String;
    public function method1():void {}
}
```

借助于点语法，可以使用在如下代码中创建的实例名来访问 prop1 属性和 method1() 方法：

```
var myDotEx:DotExample = new DotExample();
myDotEx.prop1 = "hello";
myDotEx.method1();
```

定义包时，可以使用点语法。可以使用点运算符来引用嵌套包。例如，EventDispatcher 类位于一个名为 events 的包中，该包嵌套在名为 flash 的包中。可以使用下面的表达式来引用 events 包：

```
flash.events
```

还可以使用此表达式来引用 EventDispatcher 类：

```
flash.events.EventDispatcher
```

斜杠语法

ActionScript 3.0 不支持斜杠语法。在早期的 ActionScript 版本中，斜杠语法用于指示影片剪辑或变量的路径。

字面值

文本是直接出现在代码中的值。下面的例子都是文本：

```
17
"hello"
-3
9.4
null
undefined
true
false
```

字面值还可以组合起来构成复合字面值。数组文本括在中括号字符 ([]) 中，各数组元素之间用逗号隔开。

数组文本可用于初始化一个数组。下例显示了使用数组文本进行初始化的两个数组。您可以使用 new 语句将复合字面值作为参数传递给 Array 类构造函数，但是，您还可以在实例化下面的 ActionScript 核心类的实例时直接赋予字面值：Object、Array、String、Number、int、uint、XML、XMLList 和 Boolean。

```
// Use new statement.
var myStrings:Array = new Array(["alpha", "beta", "gamma"]);
var myNums:Array = new Array([1,2,3,5,8]);

// Assign literal directly.
var myStrings:Array = ["alpha", "beta", "gamma"];
var myNums:Array = [1,2,3,5,8];
```

字面值还可用来初始化通用对象。通用对象是对象类的一个实例。对象字面值括在大括号 ({}) 中，各对象属性之间用逗号隔开。每个属性都用冒号字符 (:) 进行声明，冒号用于分隔属性名和属性值。

可以使用 new 语句创建一个通用对象并将该对象的字面值作为参数传递给 Object 类构造函数，也可以在声明实例时直接将对象字面值赋给实例。下面的示例演示两个替代方法，用于创建一个新的通用对象，并使用三个值分别设置为 1、2 和 3 的属性 (propA、propB 和 propC) 初始化该对象：

```
// Use new statement and add properties.
var myObject:Object = new Object();
myObject.propA = 1;
myObject.propB = 2;
myObject.propC = 3;

// Assign literal directly.
var myObject:Object = {propA:1, propB:2, propC:3};
```

有关详细信息，请参阅第 116 页的“[字符串基础知识](#)”、第 171 页的“[正则表达式基础知识](#)”和第 195 页的“[初始化 XML 变量](#)”。

分号

可以使用分号字符 (;) 来终止语句。如果您省略分号字符，则编译器将假设每一行代码代表一条语句。由于很多程序员都习惯使用分号来表示语句结束，因此，如果您坚持使用分号来终止语句，则代码会更易于阅读。

使用分号终止语句可以在一行中放置多个语句，但是这样会使代码变得难以阅读。

小括号

在 ActionScript 3.0 中，可以通过三种方式使用小括号 (())。首先，可以使用小括号来更改表达式中的运算顺序。组合到小括号中的运算总是最先执行。例如，小括号可用来改变如下代码中的运算顺序：

```
trace(2 + 3 * 4); // 14
trace((2 + 3) * 4); // 20
```

第二，可以结合使用小括号和逗号运算符 (,) 来计算一系列表达式并返回最后一个表达式的结果，如下面的示例所示：

```
var a:int = 2;
var b:int = 3;
trace((a++, b++, a+b)); // 7
```

第三，可以使用小括号来向函数或方法传递一个或多个参数，如下面的示例所示，此示例向 `trace()` 函数传递一个字符串值：

```
trace("hello"); // hello
```

注释

ActionScript 3.0 代码支持两种类型的注释：单行注释和多行注释。这些注释机制与 C++ 和 Java 中的注释机制类似。编译器将忽略标记为注释的文本。

单行注释以两个正斜杠字符 (`//`) 开头并持续到该行的末尾。例如，下面的代码包含一个单行注释：

```
var someNumber:Number = 3; // a single line comment
```

多行注释以一个正斜杠和一个星号 (`/*`) 开头，以一个星号和一个正斜杠 (`*/`) 结尾。

```
/* This is multiline comment that can span
more than one line of code. */
```

关键字和保留字

保留字是一些单词，因为这些单词是保留给 ActionScript 使用的，所以不能在代码中将它们用作标识符。保留字包括词汇关键字，编译器将词汇关键字从程序的命名空间中移除。如果您将词汇关键字用作标识符，则编译器会报告一个错误。下表列出了 ActionScript 3.0 词汇关键字：

as	break	case	catch
class	const	continue	default
delete	do	else	extends
false	finally	for	function
if	implements	import	in
instanceof	interface	internal	is
native	new	null	package
private	protected	public	return
super	switch	this	throw
to	true	try	typeof
use	var	void	while
with			

有一小组称为句法关键字的关键字，这些关键字可用作标识符，但是在某些上下文中具有特殊的含义。下表列出了 ActionScript 3.0 句法关键字：

each	get	set	namespace
include	dynamic	final	native
override	static		

还有几个有时称为供将来使用的保留字的标识符。这些标识符不是为 ActionScript 3.0 保留的，但是其中的一些可能会被采用 ActionScript 3.0 的软件视为关键字。可以在您的代码中使用其中的许多标识符，但是 Adobe 建议您不要使用这些标识符，因为它们可能在未来版本的语言中作为关键字。

abstract	boolean	byte	cast
char	debugger	double	enum
export	float	goto	intrinsic
long	prototype	short	synchronized
throws	to	transient	type
virtual	volatile		

常数

ActionScript 3.0 支持 `const` 语句，该语句可用来创建常量。常量是指具有无法改变的固定值的属性。只能为常量赋值一次，而且必须在最接近常量声明的位置赋值。例如，如果将常量声明为类的成员，则只能在声明过程中或者在类构造函数中为常量赋值。

下面的代码声明两个常量。第一个常量 `MINIMUM` 是在声明语句中赋值的，第二个常量 `MAXIMUM` 是在构造函数中赋值的。请注意，此示例仅在标准模式下进行编译，因为严格模式只允许在初始化时对常量进行赋值。

```
class A
{
    public const MINIMUM:int = 0;
    public const MAXIMUM:int;

    public function A()
    {
        MAXIMUM = 10;
    }
}

var a:A = new A();
trace(a.MINIMUM); // 0
trace(a.MAXIMUM); // 10
```

如果您尝试以其它任何方法向常量赋予初始值，则会出现错误。例如，如果您尝试在类的外部设置 `MAXIMUM` 的初始值，将会出现运行时错误。

```
class A
{
    public const MINIMUM:int = 0;
    public const MAXIMUM:int;
}

var a:A = new A();
a["MAXIMUM"] = 10; // run-time error
```

ActionScript 3.0 定义了各种各样的常量供您使用。按照惯例，ActionScript 中的常量全部使用大写字母，各个单词之间用下划线字符（_）分隔。例如，`MouseEvent` 类定义将此命名惯例用于其常量，其中每个常量都表示一个与鼠标输入有关的事件：

```

package flash.events
{
    public class MouseEvent extends Event
    {
        public static const CLICK:String = "click";
        public static const DOUBLE_CLICK:String = "doubleClick";
        public static const MOUSE_DOWN:String = "mouseDown";
        public static const MOUSE_MOVE:String = "mouseMove";
        ...
    }
}

```

运算符

运算符是一种特殊的函数，它们具有一个或多个操作数并返回相应的值。操作数是运算符用作输入的值（通常为字面值、变量或表达式）。例如，在下面的代码中，将加法运算符（+）和乘法运算符（*）与三个字面值操作数（2、3和4）结合使用来返回一个值。赋值运算符（=）随后使用此值将返回值14赋给变量sumNumber。

```
var sumNumber:uint = 2 + 3 * 4; // uint = 14
```

运算符可以是一元、二元或三元的。一元运算符采用1个操作数。例如，递增运算符（++）就是一元运算符，因为它只有一个操作数。二元运算符采用2个操作数。例如，除法运算符（/）有2个操作数。三元运算符采用3个操作数。例如，条件运算符（?:）采用3个操作数。

有些运算符是重载的，这意味着其行为因传递给它们的操作数的类型或数量而异。例如，加法运算符（+）就是一个重载运算符，其行为因操作数的数据类型而异。如果两个操作数都是数字，则加法运算符会返回这些值的和。如果两个操作数都是字符串，则加法运算符会返回这两个操作数连接后的结果。下面的示例代码说明运算符的行为如何因操作数而异：

```
trace(5 + 5); // 10
trace("5" + "5"); // 55
```

运算符的行为还可能因所提供的操作数的数量而异。减法运算符（-）既是一元运算符又是二元运算符。对于减法运算符，如果只提供一个操作数，则该运算符会对操作数求反并返回结果；如果提供两个操作数，则减法运算符返回这两个操作数的差。下面的示例说明首先将减法运算符用作一元运算符，然后再将其用作二元运算符。

```
trace(-3); // -3
trace(7 - 2); // 5
```

运算符的优先级和结合律

运算符的优先级和结合律决定了处理运算符的顺序。虽然对于熟悉算术的人来说，编译器先处理乘法运算符（*）然后再处理加法运算符（+）似乎是自然而然的事情，但实际上编译器要求显式指定先处理哪些运算符。此类指令统称为运算符优先级。

ActionScript 定义了一个默认的运算符优先级，可以使用小括号运算符（()）来改变它。例如，下面的代码改变上一个示例中的默认优先级，以强制编译器先处理加法运算符，然后再处理乘法运算符：

```
var sumNumber:uint = (2 + 3) * 4; // uint == 20
```

您可能会遇到这样的情况：同一个表达式中出现两个或更多个具有相同的优先级的运算符。在这些情况下，编译器使用结合律的规则确定首先处理哪个运算符。除了赋值运算符之外，所有二进制运算符中都是左结合的，也就是说，先处理左边的运算符然后再处理右边的运算符。赋值运算符和条件运算符（?:）都是右结合的，也就是说先处理右边的运算符然后再处理左边的运算符。

例如，考虑小于运算符（<）和大于运算符（>），这两个运算符具有相同的优先级。如果将这两个运算符用于同一个表达式中，因为这两个运算符都是左结合的，所以首先处理左边的运算符。也就是说，以下两个语句将生成相同的输出结果：

```
trace(3 > 2 < 1); // false
trace((3 > 2) < 1); // false
```

会首先处理大于运算符，这会生成值 `true`，因为操作数 3 大于操作数 2。随后，将值 `true` 与操作数 1 一起传递给小于运算符。下面的代码表示此中间状态：

```
trace((true) < 1);
```

小于运算符将值 `true` 转换为数值 1，然后将该数值与第二个操作数 1 进行比较，这会返回值 `false`（因为值 1 不小于 1）。

```
trace(1 < 1); // false
```

您可以用括号运算符来改变默认的左结合律。您可以通过用小括号括起小于运算符及其操作数来命令编译器先处理小于运算符。下面的示例使用与上一个示例相同的数，但是因为使用了小括号运算符，所以生成不同的输出结果：

```
trace(3 > (2 < 1)); // true
```

将首先处理小于运算符，这会生成值 `false`，因为操作数 2 不小于操作数 1。值 `false` 随后将与操作数 3 一起传递给大于运算符。

下面的代码表示此中间状态：

```
trace(3 > (false));
```

大于运算符将值 `false` 转换为数值 0，然后将该数值与另一个操作数 3 进行比较，这会返回 `true`（因为 3 大于 0）。

```
trace(3 > 0); // true
```

下表按优先级递减的顺序列出了 ActionScript 3.0 中的运算符。在该表中，每一行中包含的运算符优先级相同。表中每一行运算符的优先级都高于出现在它下面的行中的运算符。

组合	运算符
主要	<code>[] {x:y} () f(x) new x.y x[y] <></> @ :: ..</code>
后缀	<code>x++ x--</code>
一元	<code>++x --x + - ~ ! delete typeof void</code>
乘法	<code>* / %</code>
加法	<code>+ -</code>
按位移位	<code><< >> >>></code>
关系	<code>< > <= >= as in instanceof is</code>
等于	<code>== != === !==</code>
按位“与”	<code>&</code>
按位“异或”	<code>^</code>
按位“或”	<code> </code>
逻辑“与”	<code>&&</code>
逻辑“或”	<code> </code>
条件	<code>? :</code>
赋值	<code>= *= /= %= += -= <<= >>= >>>= &= ^= =</code>
逗号	<code>,</code>

主要运算符

主要运算符包括那些用来创建 `Array` 和 `Object` 字面值、对表达式进行分组、调用函数、实例化类实例以及访问属性的运算符。

下表列出了所有主要运算符，它们具有相同的优先级。属于 E4X 规范的运算符用 (E4X) 来表示。

运算符	执行的运算
[]	初始化数组
{x:y}	初始化对象
()	对表达式进行分组
f(x)	调用函数
new	调用构造函数
x.y x[y]	访问属性
<></>	初始化 XMLList 对象 (E4X)
@	访问属性 (E4X)
::	限定名称 (E4X)
..	访问子级 XML 元素 (E4X)

后缀运算符

后缀运算符只有一个操作数，它递增或递减该操作数的值。虽然这些运算符是一元运算符，但是它们有别于其它一元运算符，被单独划归到了一个类别，因为它们具有更高的优先级和特殊的行为。在将后缀运算符用作较长表达式的一部分时，先返回表达式的值后处理后缀运算符。例如，下面的代码演示了如何在值递增之前返回表达式 xNum++ 的值。

```
var xNum:Number = 0;
trace(xNum++); // 0
trace(xNum); // 1
```

下表列出了所有的后缀运算符，它们具有相同的优先级：

运算符	执行的运算
++	递增 (后缀)
--	递减 (后缀)

一元运算符

一元运算符只有一个操作数。这一组中的递增运算符 (++) 和递减运算符 (--) 是前缀运算符，这意味着它们在表达式中出现在操作数的前面。前缀运算符与它们对应的后缀运算符不同，因为递增或递减操作是在返回整个表达式的值之前完成的。例如，下面的代码演示如何在递增值之后返回表达式 ++xNum 的值：

```
var xNum:Number = 0;
trace(++xNum); // 1
trace(xNum); // 1
```

下表列出了所有的一元运算符，它们具有相同的优先级：

运算符	执行的运算
++	递增 (前缀)
--	递减 (前缀)
+	一元 +
-	一元 - (非)
!	逻辑“非”

运算符	执行的运算
<code>~</code>	按位“非”
<code>delete</code>	删除属性
<code>typeof</code>	返回类型信息
<code>void</code>	返回未定义值

乘法运算符

乘法运算符有两个操作数，执行乘、除或求模计算。

下表列出了所有的乘法运算符，它们具有相同的优先级：

运算符	执行的运算
<code>*</code>	乘法
<code>/</code>	除法
<code>%</code>	求模

加法运算符

加法运算符有两个操作数，分别执行加法或减法计算。下表列出了所有加法运算符，它们具有相同的优先级：

运算符	执行的运算
<code>+</code>	加法
<code>-</code>	减法

按位移位运算符

按位移位运算符有两个操作数，将第一个操作数的各位按第二个操作数指定的长度移位。下表列出了所有按位移位运算符，它们具有相同的优先级：

运算符	执行的运算
<code><<</code>	按位左移位
<code>>></code>	按位右移位
<code>>>></code>	按位无符号右移位

关系运算符

关系运算符有两个操作数，它比较两个操作数的值，然后返回一个布尔值。下表列出了所有关系运算符，它们具有相同的优先级：

运算符	执行的运算
<code><</code>	小于
<code>></code>	大于
<code><=</code>	小于或等于
<code>>=</code>	大于或等于

运算符	执行的运算
as	检查数据类型
in	检查对象属性
instanceof	检查原型链
is	检查数据类型

等于运算符

等于运算符有两个操作数，它比较两个操作数的值，然后返回一个布尔值。下表列出了所有等于运算符，它们具有相同的优先级：

运算符	执行的运算
==	等于
!=	不等于
====	全等
!==	不全等

按位逻辑运算符

按位逻辑运算符有两个操作数，执行位级别的逻辑运算。按位逻辑运算符具有不同的优先级；下表按优先级递减的顺序列出了按位逻辑运算符：

运算符	执行的运算
&	按位“与”
^	按位“异或”
	按位“或”

逻辑运算符

逻辑运算符有两个操作数，这类运算符返回布尔结果。逻辑运算符具有不同的优先级；下表按优先级递减的顺序列出了逻辑运算符：

运算符	执行的运算
&&	逻辑“与”
	逻辑“或”

条件运算符

条件运算符是一个三元运算符，也就是说它有三个操作数。条件运算符是应用 if..else 条件语句的一种简化方式。

运算符	执行的运算
? :	条件

赋值运算符

赋值运算符有两个操作数，它根据一个操作数的值对另一个操作数进行赋值。下表列出了所有赋值运算符，它们具有相同的优先级：

运算符	执行的运算
=	赋值
*=	乘法赋值
/=	除法赋值
%=	求模赋值
+=	加法赋值
-=	减法赋值
<<=	按位左移位赋值
>>=	按位右移位赋值
>>>=	按位无符号右移位赋值
&=	按位“与”赋值
^=	按位“异或”赋值
=	按位“或”赋值

条件语句

ActionScript 3.0 提供了三个可用来控制程序流的基本条件语句。

if..else

使用 if..else 条件语句可以测试一个条件，如果该条件存在，则执行一个代码块，如果该条件不存在，则执行替代代码块。例如，下面的代码测试 x 的值是否超过 20，如果是，则生成一个 trace() 函数，如果不是则生成另一个 trace() 函数：

```
if (x > 20)
{
    trace("x is > 20");
}
else
{
    trace("x is <= 20");
}
```

如果您不想执行替代代码块，则可以仅使用 if 语句，而不用 else 语句。

if..else if

可以使用 if..else if 条件语句测试多个条件。例如，下面的代码不仅测试 x 的值是否超过 20，而且还测试 x 的值是否为负数：

```
if (x > 20)
{
    trace("x is > 20");
}
else if (x < 0)
{
    trace("x is negative");
}
```

如果 `if` 或 `else` 语句后面只有一条语句，则无需用大括号括起该语句。例如，下面的代码不使用大括号：

```
if (x > 0)
    trace("x is positive");
else if (x < 0)
    trace("x is negative");
else
    trace("x is 0");
```

但是，Adobe 建议您始终使用大括号，因为以后在缺少大括号的条件语句中添加语句时，可能会出现意外的行为。例如，在下面的代码中，无论条件的计算结果是否为 `true`，`positiveNums` 的值总是按 1 递增：

```
var x:int;
var positiveNums:int = 0;

if (x > 0)
    trace("x is positive");
    positiveNums++;

trace(positiveNums); // 1
```

switch

如果多个执行路径依赖于同一个条件表达式，则 `switch` 语句非常有用。该语句的功能与一段 `if..else if` 系列语句类似，但是更易于阅读。`switch` 语句不是对条件进行测试以获得布尔值，而是对表达式进行求值并使用计算结果来确定要执行的代码块。代码块以 `case` 语句开头，以 `break` 语句结尾。例如，下面的 `switch` 语句基于由 `Date.getDay()` 方法返回的日期值输出星期日期：

```

var someDate:Date = new Date();
var dayNum:uint = someDate.getDay();
switch(dayNum)
{
    case 0:
        trace("Sunday");
        break;
    case 1:
        trace("Monday");
        break;
    case 2:
        trace("Tuesday");
        break;
    case 3:
        trace("Wednesday");
        break;
    case 4:
        trace("Thursday");
        break;
    case 5:
        trace("Friday");
        break;
    case 6:
        trace("Saturday");
        break;
    default:
        trace("Out of range");
        break;
}

```

循环

循环语句允许您使用一系列值或变量来反复执行一个特定的代码块。Adobe 建议您始终用大括号({})来括起代码块。尽管您可以在代码块中只包含一条语句时省略大括号，但是就像在介绍条件语言时所提到的那样，不建议您这样做，原因也相同：因为这会增加无意中将以后添加的语句从代码块中排除的可能性。如果您以后添加一条语句，并希望将它包括在代码块中，但是忘了加必要的大括号，则该语句将不会在循环过程中执行。

for

使用 for 循环可以循环访问某个变量以获得特定范围的值。必须在 for 语句中提供 3 个表达式：一个设置了初始值的变量，一个用于确定循环何时结束的条件语句，以及一个在每次循环中都更改变量值的表达式。例如，下面的代码循环 5 次。变量 i 的值从 0 开始到 4 结束，输出结果是从 0 到 4 的 5 个数字，每个数字各占 1 行。

```

var i:int;
for (i = 0; i < 5; i++)
{
    trace(i);
}

```

for..in

for..in 循环访问对象属性或数组元素。例如，可以使用 for..in 循环来循环访问通用对象的属性（不按任何特定的顺序来保存对象的属性，因此属性可能以看似随机的顺序出现）：

```

var myObj:Object = {x:20, y:30};
for (var i:String in myObj)
{
    trace(i + " : " + myObj[i]);
}
// output:
// x: 20
// y: 30

```

还可以循环访问数组中的元素：

```

var myArray:Array = ["one", "two", "three"];
for (var i:String in myArray)
{
    trace(myArray[i]);
}
// output:
// one
// two
// three

```

如果对象是自定义类的一个实例，则除非该类是动态类，否则将无法循环访问该对象的属性。即便对于动态类的实例，也只能循环访问动态添加的属性。

for each..in

for each..in 循环用于循环访问集合中的项，这些项可以是 XML 或 XMLList 对象中的标签、对象属性保存的值或数组元素。如下面这段摘录的代码所示，可以使用 for each..in 循环来循环访问通用对象的属性，但是与 for..in 循环不同的是，for each..in 循环中的迭代变量包含属性所保存的值，而不包含属性的名称：

```

var myObj:Object = {x:20, y:30};
for each (var num in myObj)
{
    trace(num);
}
// output:
// 20
// 30

```

您可以循环访问 XML 或 XMLList 对象，如下面的示例所示：

```

var myXML:XML = <users>
    <fname>Jane</fname>
    <fname>Susan</fname>
    <fname>John</fname>
</users>

for each (var item in myXML.fname)
{
    trace(item);
}
/* output
Jane
Susan
John
*/

```

还可以循环访问数组中的元素，如下面的示例所示：

```

var myArray:Array = ["one", "two", "three"];
for each (var item in myArray)
{
    trace(item);
}
// output:
// one
// two
// three

```

如果对象是密封类的实例，则您将无法循环访问该对象的属性。即使对于动态类的实例，也无法循环访问任何固定属性（即，作为类定义的一部分定义的属性）。

while

`while` 循环与 `if` 语句相似，只要条件为 `true`，就会反复执行。例如，下面的代码与 `for` 循环示例生成的输出结果相同：

```

var i:int = 0;
while (i < 5)
{
    trace(i);
    i++;
}

```

使用 `while` 循环（而非 `for` 循环）的一个缺点是，编写 `while` 循环更容易导致无限循环。如果遗漏递增计数器变量的表达式，则 `for` 循环示例代码将无法编译；而 `while` 循环示例代码将能够编译。若没有用来递增 `i` 的表达式，循环将成为无限循环。

do..while

`do..while` 循环是一种 `while` 循环，保证至少执行一次代码块，这是因为在执行代码块后才会检查条件。下面的代码显示了 `do..while` 循环的一个简单示例，该示例在条件不满足时也会生成输出结果：

```

var i:int = 5;
do
{
    trace(i);
    i++;
} while (i < 5);
// output: 5

```

函数

“函数”是执行特定任务并可以在程序中重用的代码块。`ActionScript 3.0` 中有两种函数类型：方法和函数闭包。将函数称为方法还是函数闭包取决于定义函数的上下文。如果您将函数定义为类定义的一部分或者将它附加到对象的实例，则该函数称为方法。如果您以其它任何方式定义函数，则该函数称为函数闭包。

函数在 `ActionScript` 中始终扮演着极为重要的角色。例如，在 `ActionScript 1.0` 中，不存在 `class` 关键字，因此“类”由构造函数定义。尽管 `class` 关键字已经添加到了之后的 `ActionScript` 版本中，但是，如果您想充分利用该语言所提供的功能，深入了解函数仍然十分重要。对于希望 `ActionScript` 函数的行为与 `C++` 或 `Java` 等语言中的函数的行为相似的程序员来说，这可能是一个挑战。尽管基本的函数定义和调用对有经验的程序员来说应不是什么问题，但是仍需要对 `ActionScript` 函数的一些更高级功能进行解释。

函数的基本概念

本节讨论基本的函数定义和调用方法。

调用函数

可通过使用后跟小括号运算符 () 的函数标识符来调用函数。要发送给函数的任何函数参数都括在小括号中。例如，贯穿于本书始末的 `trace()` 函数就是 ActionScript 3.0 中的顶级函数：

```
trace("Use trace to help debug your script");
```

如果要调用没有参数的函数，则必须使用一对空的小括号。例如，可以使用没有参数的 `Math.random()` 方法来生成一个随机数：

```
var randomNum:Number = Math.random();
```

定义您自己的函数

在 ActionScript 3.0 中可通过两种方法来定义函数：使用函数语句和使用函数表达式。您可以根据自己的编程风格（偏于静态还是偏于动态）来选择相应的方法。如果您倾向于采用静态或严格模式的编程，则应使用函数语句来定义函数。如果您有特定的需求，需要用函数表达式来定义函数，则应这样做。函数表达式更多地用在动态编程或标准模式编程中。

函数语句

函数语句是在严格模式下定义函数的首选方法。函数语句以 `function` 关键字开头，后跟：

- 函数名
- 用小括号括起来的逗号分隔参数列表
- 用大括号括起来的函数体 — 即，在调用函数时要执行的 ActionScript 代码

例如，下面的代码创建一个定义一个参数的函数，然后将字符串 “hello” 用作参数值来调用该函数：

```
function traceParameter(aParam:String)
{
    trace(aParam);
}

traceParameter("hello"); // hello
```

函数表达式

声明函数的第二种方法就是结合使用赋值语句和函数表达式，函数表达式有时也称为函数字面值或匿名函数。这是一种较为复杂的方法，在早期的 ActionScript 版本中广为使用。

带有函数表达式的赋值语句以 `var` 关键字开头，后跟：

- 函数名
- 冒号运算符 (:)
- 指示数据类型的 `Function` 类
- 赋值运算符 (=)
- `function` 关键字
- 用小括号括起来的逗号分隔参数列表
- 用大括号括起来的函数体 — 即，在调用函数时要执行的 ActionScript 代码

例如，下面的代码使用函数表达式来声明 `traceParameter` 函数：

```
var traceParameter:Function = function (aParam:String)
{
    trace(aParam);
};

traceParameter("hello"); // hello
```

请注意，就像在函数语句中一样，在上面的代码中也没有指定函数名。函数表达式和函数语句的另一个重要区别是，函数表达式是表达式，而不是语句。这意味着函数表达式不能独立存在，而函数语句则可以。函数表达式只能用作语句（通常是赋值语句）的一部分。下面的示例显示了一个赋予数组元素的函数表达式：

```
var traceArray:Array = new Array();
traceArray[0] = function (aParam:String)
{
    trace(aParam);
};
traceArray[0] ("hello");
```

在函数语句和函数表达式之间进行选择

原则上，除非在特殊情况下要求使用表达式，否则应使用函数语句。函数语句较为简洁，而且与函数表达式相比，更有助于保持严格模式和标准模式的一致性。

函数语句比包含函数表达式的赋值语句更便于阅读。与函数表达式相比，函数语句使代码更为简洁而且不容易引起混淆，因为函数表达式既需要 `var` 关键字又需要 `function` 关键字。

函数语句更有助于保持严格模式和标准模式的一致性，因为在这两种编译器模式下，均可以借助点语法来调用使用函数语句声明的方法。但这对于用函数表达式声明的方法却不一定成立。例如，下面的代码定义了一个具有两个方法的 `Example` 类：

`methodExpression()`（用函数表达式声明）和 `methodStatement()`（用函数语句声明）。在严格模式下，不能使用点语法来调用 `methodExpression()` 方法。

```
class Example
{
var methodExpression = function() {}
function methodStatement() {}
}

var myEx:Example = new Example();
myEx.methodExpression(); // error in strict mode; okay in standard mode
myEx.methodStatement(); // okay in strict and standard modes
```

一般认为，函数表达式更适合于关注运行时行为或动态行为的编程。如果您喜欢使用严格模式，但是还需要调用使用函数表达式声明的方法，则可以使用这两种方法中的任一方法。首先，可以使用中括号 (`[]`) 代替点运算符 (`.`) 来调用该方法。下面的方法调用在严格模式和标准模式下都能够成功执行：

```
myExample["methodLiteral"]();
```

第二，您可以将整个类声明为动态类。尽管这样您就可以使用点运算符来调用方法，但缺点是，该类的所有实例在严格模式下都将丢失一些功能。例如，如果您尝试访问动态类实例的未定义属性，则编译器不生成错误。

在某些情况下，函数表达式非常有用。函数表达式的一个常见用法就是用于那些使用一次后便丢弃的函数。另一个用法就是向原型属性附加函数，这个用法不太常见。有关详细信息，请参阅第 99 页的“[原型对象](#)”。

函数语句与函数表达式之间有两个细微的区别，在选择要使用的方法时，应考虑这两个区别。第一个区别体现在内存管理和垃圾回收方面，因为函数表达式不像对象那样独立存在。换言之，当您将某个函数表达式分配给另一个对象（如数组元素或对象属性）时，就会在代码中创建对该函数表达式的唯一引用。如果该函数表达式所附加到的数组或对象脱离作用域或由于其它原因不再可用，您将无法再访问该函数表达式。如果删除该数组或对象，该函数表达式所使用的内存将符合垃圾回收条件，这意味着内存符合回收条件并且可重新用于其它用途。

下面的示例说明对于函数表达式，一旦删除该表达式所赋予的属性，该函数就不再可用。`Test` 类是动态的，这意味着您可以添加一个名为 `functionExp` 的属性来保存函数表达式。`functionExp()` 函数可以用点运算符来调用，但是一旦删除了 `functionExp` 属性，就无法再访问该函数。

```
dynamic class Test {}
var myTest:Test = new Test();

// function expression
myTest.functionExp = function () { trace("Function expression") };
myTest.functionExp(); // Function expression
delete myTest.functionExp;
myTest.functionExp(); // error
```

另一方面，如果该函数最初是用函数语句定义的，那么，该函数将以对象的形式独立存在，即使在您删除它所附加到的属性之后，该函数仍将存在。`delete` 运算符仅适用于对象的属性，因此，即使是用于删除 `stateFunc()` 函数本身的调用也不工作。

```
dynamic class Test {}
var myTest:Test = new Test();

// function statement
function stateFunc() { trace("Function statement") }
myTest.stateFunc = stateFunc;
myTest.stateFunc(); // Function statement
delete myTest.stateFunc;
delete stateFunc; // no effect
stateFunc(); // Function statement
myTest.stateFunc(); // error
```

函数语句与函数表达式之间的第二个区别是，函数语句存在于定义它们的整个作用域（包括出现在该函数语句前面的语句）内。与之相反，函数表达式只是为后续的语句定义的。例如，下面的代码能够在定义 `scopeTest()` 函数之前成功调用该函数：

```
scopeTest(); // scopeTest

function scopeTest():void
{
    trace("scopeTest");
}
```

函数表达式只有在定义之后才可用，因此，下面的代码会生成运行时错误：

```
expressionTest(); // run-time error

var expressionTest:Function = function ()
{
    trace("expressionTest");
}
```

从函数中返回值

要从函数中返回值，请使用后跟要返回的表达式或字面值的 `return` 语句。例如，下面的代码返回一个表示参数的表达式：

```
function doubleNum(baseNum:int):int
{
    return (baseNum * 2);
}
```

请注意，`return` 语句会终止该函数，因此，不会执行位于 `return` 语句下面的任何语句，如下所示：

```
function doubleNum(baseNum:int):int {
    return (baseNum * 2);
    trace("after return"); // This trace statement will not be executed.
}
```

在严格模式下，如果您选择指定返回类型，则必须返回相应类型的值。例如，下面的代码在严格模式下会生成错误，因为它们不返回有效值：

```
function doubleNum(baseNum:int):int
{
    trace("after return");
}
```

嵌套函数

您可以嵌套函数，这意味着函数可以在其它函数内部声明。除非将对嵌套函数的引用传递给外部代码，否则嵌套函数将仅在其父函数内可用。例如，下面的代码在 `getNameAndVersion()` 函数内部声明两个嵌套函数：

```
function getNameAndVersion():String
{
    function getVersion():String
    {
        return "10";
    }
    function getProductName():String
    {
        return "Flash Player";
    }
    return (getProductName() + " " + getVersion());
}
trace(getNameAndVersion()); // Flash Player 10
```

在将嵌套函数传递给外部代码时，它们将作为函数闭包传递，这意味着嵌套函数保留在定义该函数时处于作用域内的任何定义。有关详细信息，请参阅第 73 页的“[函数作用域](#)”。

函数参数

ActionScript 3.0 为函数参数提供了一些功能，这些功能对于那些刚接触 ActionScript 语言的程序员来说可能是很陌生的。尽管大多数程序员都应熟悉按值或按引用传递参数这一概念，但是很多人可能对 `arguments` 对象和 ... (rest) 参数感到很陌生。

按值或按引用传递参数

在许多编程语言中，一定要了解按值传递参数与按引用传递参数之间的区别；这种区别会影响代码的设计方式。

按值传递意味着将参数的值复制到局部变量中以便在函数内使用。按引用传递意味着将只传递对参数的引用，而不传递实际值。这种方式的传递不会创建实际参数的任何副本，而是会创建一个对变量的引用并将它作为参数传递，并且会将它赋给局部变量以便在函数内部使用。局部变量是对函数外部的变量的引用，它使您能够更改初始变量的值。

在 ActionScript 3.0 中，所有的参数均按引用传递，因为所有的值都存储为对象。但是，属于基元数据类型（包括 Boolean、Number、int、uint 和 String）的对象具有一些特殊运算符，这使它们可以像按值传递一样工作。例如，下面的代码创建一个名为 `passPrimitives()` 的函数，该函数定义了两个类型均为 int、名称分别为 `xParam` 和 `yParam` 的参数。这些参数与在 `passPrimitives()` 函数体内声明的局部变量类似。当使用 `xValue` 和 `yValue` 参数调用函数时，`xParam` 和 `yParam` 参数将用对 int 对象的引用进行初始化，int 对象由 `xValue` 和 `yValue` 表示。因为参数是基元值，所以它们像按值传递一样工作。尽管 `xParam` 和 `yParam` 最初仅包含对 `xValue` 和 `yValue` 对象的引用，但是，对函数体内的变量的任何更改都会导致在内存中生成这些值的新副本。

```
function passPrimitives(xParam:int, yParam:int):void
{
    xParam++;
    yParam++;
    trace(xParam, yParam);
}

var xValue:int = 10;
var yValue:int = 15;
trace(xValue, yValue); // 10 15
passPrimitives(xValue, yValue); // 11 16
trace(xValue, yValue); // 10 15
```

在 `passPrimitives()` 函数内部，`xParam` 和 `yParam` 的值递增，但这不会影响 `xValue` 和 `yValue` 的值，如上一条 `trace` 语句所示。即使参数的命名与 `xValue` 和 `yValue` 变量的命名完全相同也是如此，因为函数内部的 `xValue` 和 `yValue` 将指向内存中的新位置，这些位置不同于函数外部同名的变量所在的位置。

其它所有对象（即不属于基元数据类型的对象）始终按引用传递，这样您就可以更改初始变量的值。例如，下面的代码创建一个名为 `objVar` 的对象，该对象具有两个属性：`x` 和 `y`。该对象作为参数传递给 `passByRef()` 函数。因为该对象不是基元类型，所以它不但按引用传递，而且还保持一个引用。这意味着对函数内部的参数的更改将会影响到函数外部的对象属性。

```
function passByRef(objParam:Object):void
{
    objParam.x++;
    objParam.y++;
    trace(objParam.x, objParam.y);
}
var objVar:Object = {x:10, y:15};
trace(objVar.x, objVar.y); // 10 15
passByRef(objVar); // 11 16
trace(objVar.x, objVar.y); // 11 16
```

`objParam` 参数与全局 `objVar` 变量引用相同的对象。正如在本示例的 `trace` 语句中所看到的一样，对 `objParam` 对象的 `x` 和 `y` 属性所做的更改将反映在 `objVar` 对象中。

默认参数值

ActionScript 3.0 中新增了为函数声明默认参数值的功能。如果在调用具有默认参数值的函数时省略了具有默认值的参数，那么，将使用在函数定义中为该参数指定的值。所有具有默认值的参数都必须放在参数列表的末尾。指定为默认值的值必须是编译时常量。如果某个参数存在默认值，其效果会使该参数成为可选参数。没有默认值的参数被视为必需参数。

例如，下面的代码创建一个具有三个参数的函数，其中的两个参数具有默认值。当仅用一个参数调用该函数时，将使用这些参数的默认值。

```
function defaultValues(x:int, y:int = 3, z:int = 5):void
{
    trace(x, y, z);
}
defaultValues(1); // 1 3 5
```

arguments 对象

在将参数传递给某个函数时，可以使用 `arguments` 对象来访问有关传递给该函数的参数的信息。`arguments` 对象的一些重要方面包括：

- `arguments` 对象是一个数组，其中包括传递给函数的所有参数。
- `arguments.length` 属性报告传递给函数的参数数量。
- `arguments.callee` 属性提供对函数本身的引用，该引用可用于递归调用函数表达式。

注：如果将任何参数命名为 `arguments`，或者使用 `...(rest)` 参数，则 `arguments` 对象不可用。

如果在函数体中引用了 `arguments` 对象，则在 ActionScript 3.0 中，函数调用中所包括的参数的数量可以大于在函数定义中所定义的参数数量，但是，如果参数的数量与必需参数（以及可选情况下的任何可选参数）的数量不匹配，则在严格模式下会生成编译器错误。您可以使用 `arguments` 对象的数组样式来访问传递给函数的任何参数，而无需考虑是否在函数定义中定义了该参数。下面的示例（仅在标准模式下进行编译）使用 `arguments` 数组及 `arguments.length` 属性来跟踪传递给 `traceArgArray()` 函数的所有参数：

```

function traceArgArray(x:int):void
{
    for (var i:uint = 0; i < arguments.length; i++)
    {
        trace(arguments[i]);
    }
}

traceArgArray(1, 2, 3);

// output:
// 1
// 2
// 3

```

`argumentscallee` 属性通常用在匿名函数中以创建递归。您可以使用它来提高代码的灵活性。如果递归函数的名称在开发周期内的不同阶段会发生改变，而且您使用的是 `argumentscallee`（而非函数名），则不必花费精力在函数体内更改递归调用。下面的函数表达式中使用 `argumentscallee` 属性来启用递归：

```

var factorial:Function = function (x:uint)
{
    if(x == 0)
    {
        return 1;
    }
    else
    {
        return (x * argumentscallee(x - 1));
    }
}

trace(factorial(5)); // 120

```

如果在函数声明中使用 `... (rest)` 参数，则不能使用 `arguments` 对象。而必须使用为参数声明的参数名来访问参数。

还应避免将 "arguments" 字符串作为参数名，因为该字符串会遮蔽 `arguments` 对象。例如，如果重写 `traceArgArray()` 函数，以便添加 `arguments` 参数，那么，函数体内对 `arguments` 的引用所引用的将是该参数，而不是 `arguments` 对象。下面的代码不生成输出结果：

```

function traceArgArray(x:int, arguments:int):void
{
    for (var i:uint = 0; i < arguments.length; i++)
    {
        trace(arguments[i]);
    }
}

traceArgArray(1, 2, 3);

// no output

```

在早期的 ActionScript 版本中，`arguments` 对象还包含一个名为 `caller` 的属性，该属性是对当前函数的引用。ActionScript 3.0 中没有 `caller` 属性，但是，如果您需要引用调用函数，则可以更改调用函数，以使其传递一个额外的参数来引用其本身。

... (rest) 参数

ActionScript 3.0 中引入了一个称为 `... (rest)` 参数的新参数声明。此参数可用来指定一个数组参数以接受任意多个以逗号分隔的参数。参数可以拥有保留字以外的任意名称。此参数声明必须是最后一个指定的参数。使用此参数会使 `arguments` 对象不可用。尽管 `... (rest)` 参数提供了与 `arguments` 数组和 `arguments.length` 属性相同的功能，但是未提供与由 `argumentscallee` 提供的功能类似的功能。使用 `...(rest)` 参数之前，应确保不需要使用 `argumentscallee`。

下面的示例使用 ... (rest) 参数（而不是 arguments 对象）来重写 traceArgArray() 函数：

```
function traceArgArray(... args):void
{
    for (var i:uint = 0; i < args.length; i++)
    {
        trace(args[i]);
    }
}

traceArgArray(1, 2, 3);

// output:
// 1
// 2
// 3
```

... (rest) 参数还可与其它参数一起使用，前提是 ... (rest) 参数是列出的最后一个参数。下面的示例修改 traceArgArray() 函数，以便其第一个参数 x 是 int 类型，第二个参数使用 ... (rest) 参数。输出结果会忽略第一个值，因为第一个参数不再属于由 ... (rest) 参数创建的数组。

```
function traceArgArray(x: int, ... args)
{
    for (var i:uint = 0; i < args.length; i++)
    {
        trace(args[i]);
    }
}

traceArgArray(1, 2, 3);

// output:
// 2
// 3
```

函数作为对象

ActionScript 3.0 中的函数是对象。当您创建函数时，就是在创建对象，该对象不仅可以作为参数传递给另一个函数，而且还可以有附加的属性和方法。

作为参数传递给另一个函数的函数是按引用（而不是按值）传递的。在将某个函数作为参数传递时，只能使用标识符，而不能使用在调用方法时所用的小括号运算符。例如，下面的代码将名为 clickListener() 的函数作为参数传递给 addEventListener() 方法：

```
addEventListener(MouseEvent.CLICK, clickListener);
```

Array.sort() 方法也定义了一个接受函数的参数。有关用作 Array.sort() 函数参数的自定义排序函数的示例，请参阅第 134 页的“[对数组排序](#)”。

尽管刚接触 ActionScript 的程序员可能对此感觉有些奇怪，但是，函数确实可以像其它任何对象那样具有属性和方法。实际上，每个函数都有一个名为 length 的只读属性，它用来存储为该函数定义的参数数量。该属性与 arguments.length 属性不同，后者报告发送给函数的参数数量。回想一下，在 ActionScript 中，发送给函数的参数数量可以超过为该函数定义的参数数量。下面的示例说明了这两个属性之间的区别，此示例仅在标准模式下编译，因为严格模式要求所传递的参数数量与所定义的参数数量完全相同：

```
// Compiles only in standard mode
function traceLength(x:uint, y:uint):void
{
    trace("arguments received: " + arguments.length);
    trace("arguments expected: " + traceLength.length);
}

traceLength(3, 5, 7, 11);
/* output:
arguments received: 4
arguments expected: 2 */
```

在标准模式下，可以通过在函数体外部定义函数属性来定义您自己的函数属性。函数属性可以用作准静态属性，用来保存与该函数有关的变量的状态。例如，您可能希望跟踪对特定函数的调用次数。如果您正在编写游戏，并且希望跟踪用户使用特定命令的次数，则这样的功能会非常有用，尽管您也可以使用静态类属性来实现此目的。下面的示例（该示例仅在标准模式下进行编译，因为严格模式不允许向函数添加动态属性）在函数声明外部创建一个函数属性，并在每次调用该函数时递增此属性：

```
// Compiles only in standard mode
var someFunction:Function = function ():void
{
    someFunction.counter++;
}

someFunction.counter = 0;

someFunction();
someFunction();
trace(someFunction.counter); // 2
```

函数作用域

函数的范围不但决定了可以在程序中的什么位置调用函数，而且还决定了函数可以访问哪些定义。适用于变量标识符的作用域规则同样也适用于函数标识符。在全局作用域中声明的函数在整个代码中都可用。例如，ActionScript 3.0 包含可在代码中的任意位置使用的全局函数，如 isNaN() 和 parseInt()。嵌套函数（即在另一个函数中声明的函数）可以在声明它的函数中的任意位置上使用。

作用域链

无论何时开始执行函数，都会创建许多对象和属性。首先，会创建一个称为激活对象的特殊对象，该对象用于存储在函数体内声明的参数以及任何局部变量或函数。由于激活对象属于内部机制，因此您无法直接访问它。接着，会创建一个范围链，其中包含由 Flash Player 或 Adobe AIR 检查标识符声明的对象的有序列表。所执行的每个函数都有一个存储在内部属性中的作用域链。对于嵌套函数，范围链始于其自己的激活对象，后跟其父函数的激活对象。作用域链以这种方式延伸，直到到达全局对象。全局对象是在 ActionScript 程序开始时创建的，其中包含所有的全局变量和函数。

函数闭包

函数闭包是一个对象，其中包含函数的快照及其“词汇环境”。函数的词汇环境包括函数范围链中的所有变量、属性、方法和对象以及它们的值。无论何时在对象或类之外的位置执行函数，都会创建函数闭包。函数闭包保留定义它们的作用域，这样，在将函数作为参数或返回值传递给另一个作用域时，会产生有趣的结果。

例如，下面的代码创建两个函数：foo()（返回一个用来计算矩形面积的嵌套函数 rectArea()）和 bar()（调用 foo() 并将返回的函数闭包存储在名为 myProduct 的变量中）。即使 bar() 函数定义了自己的局部变量 x（值为 2），当调用函数闭包 myProduct() 时，该函数闭包仍保留在函数 foo() 中定义的变量 x（值为 40）。因此，bar() 函数会返回值 160，而不是 8。

```
function foo():Function
{
    var x:int = 40;
    function rectArea(y:int):int // function closure defined
    {
        return x * y
    }
    return rectArea;
}
function bar():void
{
    var x:int = 2;
    var y:int = 4;
    var myProduct:Function = foo();
    trace(myProduct(4)); // function closure called
}
bar(); // 160
```

方法的行为与函数闭包类似，因为方法也保留有关创建它们的词汇环境的信息。当方法提取自它的实例（这会创建绑定方法）时，此特征尤为突出。函数闭包与绑定方法之间的主要区别在于，绑定方法中 `this` 关键字的值始终引用它最初附加到的实例，而函数闭包中 `this` 关键字的值可以改变。有关详细信息，请参阅第 81 页的“[方法](#)”。

第 5 章：ActionScript 中面向对象的编程

本章介绍支持面向对象的编程 (OOP) 的 ActionScript 元素。本章没有介绍一般 OOP 原则，如对象设计、抽象、封装、继承和多态。本章重点说明如何在使用 ActionScript 3.0 时应用这些原则。

从根本上讲，ActionScript 是一种脚本撰写语言，因此 ActionScript 3.0 OOP 支持是可选的。这为程序员提供了很大的灵活性，使他们可以为各种领域和不同复杂程度的项目选择最佳方法。对于小型任务，您可能发现使用 ActionScript 和过程化程序设计方法就能满足所有需要。对于大型项目，应用 OOP 原则可以使代码更易于理解、维护和扩展。

面向对象的编程基础知识

面向对象的编程简介

面向对象的编程 (OOP) 是一种组织程序代码的方法，它将代码划分为对象，即包含信息（数据值）和功能的单个元素。通过使用面向对象的方法来组织程序，可以将特定信息（例如，唱片标题、音轨标题或歌手名字等音乐信息）及其关联的通用功能或动作（如“在播放列表中添加音轨”或“播放此歌手的所有歌曲”）组合在一起。这些项目将合并为一个项目，即对象（例如，“唱片”或“音轨”）。能够将这些值和功能捆绑在一起会带来很多好处，其中包括只需跟踪单个变量而非多个变量、将相关功能组织在一起，以及能够以更接近实际情况的方式构建程序。

常见的面向对象编程任务

实际上，面向对象的编程包含两个部分。一部分是程序设计策略和技巧（通常称为面向对象的设计）。这是一个很广泛的主题，本章中不对其进行讨论。OOP 的另一部分是在给定编程语言中提供的实际编程结构，以便使用面向对象的方法来构建程序。本章介绍了 OOP 中的以下常见任务：

- 定义类
- 创建属性、方法以及 `get` 和 `set` 存取器（存取器方法）
- 控制对类、属性、方法和存取器的访问
- 创建静态属性和方法
- 创建与枚举类似的结构
- 定义和使用接口
- 处理继承（包括覆盖类元素）

重要概念和术语

以下参考列表包含将会在本章中遇到的重要术语：

- 属性：在类定义中为类元素（如属性或方法）分配的特性。属性通常用于定义程序的其它部分中的代码能否访问属性或方法。例如，`private` 和 `public` 都是属性。私有方法只能由类中的代码调用；而公共方法可以由程序中的任何代码调用。
- 类：某种类型的对象的结构和行为定义（与该数据类型的对象的模板或蓝图类似）。
- 类层次结构：多个相关的类的结构，用于指定哪些类继承了其它类中的功能。
- 构造函数：可以在类中定义的特殊方法，创建类的实例时将调用该方法。构造函数通常用于指定默认值，或以其它方式执行对象的设置操作。
- 数据类型：特定变量可以存储的信息类型。通常，数据类型表示与类相同的内容。

- 点运算符：句点符号(.)，在 ActionScript（和很多其它编程语言）中，它用于指示某个名称引用对象的子元素（如属性或方法）。例如，在表达式 myObject.myProperty 中，点运算符指示 myProperty 项引用的值是名为 myObject 的对象的元素。
- 枚举：一组相关常数值，为方便起见而将其作为一个类的属性组合在一起。
- 继承：一种 OOP 机制，它允许一个类定义包含另一个类定义的所有功能（通常会添加到该功能中）。
- 实例：在程序中创建的实际对象。
- 命名空间：实质上是一个自定义属性，它可以更精确地控制代码对其他代码的访问。

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。由于本章中的代码清单主要用于定义和处理数据类型，测试示例将涉及创建要定义的类的实例，使用该实例的属性或方法来处理该实例，然后查看该实例属性的值。为了查看这些值，您需要将值写入舞台上的文本字段实例，或使用 trace() 函数将值输出到“输出”面板。第 29 页的“[测试本章内的示例代码清单](#)”将详细介绍这些方法。

类

类是对象的抽象表示形式。类用来存储有关对象可保存的数据类型及对象可表现的行为的信息。如果编写的小脚本中只包含几个彼此交互的对象，使用这种抽象类的作用可能并不明显。但是，随着程序作用域不断扩大以及必须管理的对象数不断增加，您可能会发现，可以使用类更好地控制对象的创建方式以及对象之间的交互方式。

早在 ActionScript 1.0 中，ActionScript 程序员就能使用 Function 对象创建类似类的构造函数。在 ActionScript 2.0 中，通过使用 class 和 extends 等关键字，正式添加了对类的支持。ActionScript 3.0 不但继续支持 ActionScript 2.0 中引入的关键字，而且还添加了一些新功能，如通过 protected 和 internal 属性增强了访问控制，通过 final 和 override 关键字增强了对继承的控制。

如果您曾经使用类似 Java、C++ 或 C# 这样的编程语言创建过类，就会发现 ActionScript 中的实现方法与之类似。

ActionScript 共享了许多相同的关键字和属性名，如 class、extends 和 public，以下各部分将讨论所有这些内容。

注：本章中，术语“属性”表示对象或类的任何成员，包括变量、常量和方法。此外，虽然术语“类”和“静态”经常互换使用，但在本章中这两个术语的概念是不同的。例如，本章中短语“类属性”指的是类的所有成员，而不仅是静态成员。

类定义

ActionScript 3.0 类定义使用的语法与 ActionScript 2.0 类定义使用的语法相似。正确的类定义语法中要求 class 关键字后跟类名。类体要放在大括号 {} 内，且放在类名后面。例如，下面的代码创建名为 Shape 的类，其中包含名为 visible 的变量：

```
public class Shape
{
    var visible:Boolean = true;
}
```

对于包中的类定义，有一项重要的语法更改。在 ActionScript 2.0 中，如果类在包中，则在类声明中必须包含包名称。在 ActionScript 3.0 中，引入了 package 语句，包名称必须包含在包声明中，而不是包含在类声明中。例如，以下类声明说明如何在 ActionScript 2.0 和 ActionScript 3.0 中定义 BitmapData 类（该类是 flash.display 包的一部分）：

```
// ActionScript 2.0
class flash.display.BitmapData {}

// ActionScript 3.0
package flash.display
{
    public class BitmapData {}
}
```

类属性

在 ActionScript 3.0 中，可使用以下四个属性之一来修改类定义：

属性	定义
dynamic	允许在运行时向实例添加属性。
final	不得由其它类扩展。
internal (默认)	对当前包内的引用可见。
public	对所有位置的引用可见。

使用 `internal` 以外的每个属性时，必须显式包含该属性才能获得相关的行为。例如，如果定义类时未包含 `dynamic` 属性 (`attribute`)，则不能在运行时向类实例中添加属性 (`property`)。通过在类定义的开始处放置属性，可显式地分配属性，如下面的代码所示：

```
dynamic class Shape {}
```

请注意，列表中未包含名为 `abstract` 的属性。这时因为 ActionScript 3.0 不支持抽象类。另外，该清单也未包含名为 `private` 和 `protected` 的属性。这些属性只在类定义中有意义，但不可以应用于类本身。如果不希望某个类在包以外公开可见，请将该类放在包中，并用 `internal` 属性标记该类。或者，可以省略 `internal` 和 `public` 这两个属性，编译器会自动为您添加 `internal` 属性。如果不希望某个类在定义该类的源文件以外可见，请将类放在包定义右大括号下面的源文件底部。

类体

类体放在大括号内，用于定义类的变量、常量和方法。下面的示例显示 Adobe Flash Player API 中 `Accessibility` 类的声明：

```
public final class Accessibility
{
    public static function get active():Boolean;
    public static function updateProperties():void;
}
```

还可以在类体中定义命名空间。下面的示例说明如何在类体中定义命名空间，以及如何在该类中将命名空间用作方法的属性：

```
public class SampleClass
{
    public namespace sampleNamespace;
    sampleNamespace function doSomething():void;
}
```

ActionScript 3.0 不但允许在类体中包括定义，而且还允许包括语句。如果语句在类体中但在方法定义之外，这些语句只在第一次遇到类定义并且创建了相关的类对象时执行一次。下面的示例包括一个对 `hello()` 外部函数的调用和一个 `trace` 语句，该语句在定义类时输出确认消息：

```

function hello():String
{
    trace("hola");
}
class SampleClass
{
    hello();
    trace("class created");
}
// output when class is created
hola
class created

```

与以前版本的 ActionScript 相比，ActionScript 3.0 中允许在同一类体中定义同名的静态属性和实例属性。例如，下面的代码声明一个名为 message 的静态变量和一个同名的实例变量：

```

class StaticTest
{
    static var message:String = "static variable";
    var message:String = "instance variable";
}
// In your script
var myST:StaticTest = new StaticTest();
trace(StaticTest.message); // output: static variable
trace(myST.message); // output: instance variable

```

类属性 (property) 的属性 (attribute)

讨论 ActionScript 对象模型时，术语属性 指可以作为类成员的任何成员，包括变量、常量和方法。这与《ActionScript 3.0 语言和组件参考》中该术语的使用方式有所不同，后者中该术语的使用范围更窄，只包括作为变量的类成员或用 `getter` 或 `setter` 方法定义的类成员。在 ActionScript 3.0 中，提供了可以与类的任何属性 (property) 一起使用的一组属性 (attribute)。下表列出了这组属性。

属性	定义
internal (默认)	对同一包中的引用可见。
private	对同一类中的引用可见。
protected	对同一类及派生类中的引用可见。
public	对所有位置的引用可见。
static	指定某一属性属于该类，而不属于该类的实例。
UserDefinedNamespace	用户定义的自定义命名空间名。

访问控制命名空间属性

ActionScript 3.0 提供了四个特殊的属性 (attribute) 来控制对在类中定义的属性 (property) 的访问：public、private、protected 和 internal。

使用 public 属性 (attribute) 可使某一属性 (property) 在脚本的任何位置可见。例如，要使某个方法可用于包外部的代码，必须使用 public 属性声明该方法。这适用于任何属性，不管属性是使用 var、const 还是 function 关键字声明的。

使用 private 属性 (attribute) 可使某一属性 (property) 只对属性 (property) 的定义类中的调用方可见。这一行为不同于 ActionScript 2.0 中 private 属性 (attribute) 的行为，后者允许子类访问超类中的私有属性 (property)。另一处明显的行为变化是必须执行运行时访问。在 ActionScript 2.0 中，private 关键字只在编译时禁止访问，运行时很容易避开它。在 ActionScript 3.0 中，这种情况不复存在。标记为 private 的属性在编译时和运行时都不可用。

例如，下面的代码创建了名为 **PrivateExample** 的简单类，其中包含一个私有变量，然后尝试从该类的外部访问该私有变量。在 ActionScript 2.0 中，编译时访问被禁止，但是使用属性访问运算符 (`[]`) 可以很容易地避开禁止，属性访问运算符在运行时（而不是编译时）执行属性查找。

```
class PrivateExample
{
    private var privVar:String = "private variable";
}

var myExample:PrivateExample = new PrivateExample();
trace(myExample.privVar); // compile-time error in strict mode
trace(myExample["privVar"]); // ActionScript 2.0 allows access, but in ActionScript 3.0, this is a run-time error.
```

在 ActionScript 3.0 中使用严格模式时，尝试使用点运算符 (`myExample.privVar`) 访问私有属性会导致编译时错误。否则，会在运行时报告错误，就像使用属性访问运算符 (`myExample["privVar"]`) 时一样。

下表汇总了试图访问属于密封（非动态）类的 **private** 属性的结果：

	严格模式	标准模式
点运算符 ()	编译时错误	运行时错误
中括号运算符 ([])	运行时错误	运行时错误

在使用 **dynamic** 属性声明的类中尝试访问私有变量时，不会导致运行时错误。只是变量不可见，所以 Flash Player or Adobe® AIR™ 返回值 **undefined**。但是，如果在严格模式下使用点运算符，则会发生编译时错误。下面的示例与上一个示例相同，只是 **PrivateExample** 类被声明为动态类：

```
dynamic class PrivateExample
{
    private var privVar:String = "private variable";
}

var myExample:PrivateExample = new PrivateExample();
trace(myExample.privVar); // compile-time error in strict mode
trace(myExample["privVar"]); // output: undefined
```

当类外部的代码尝试访问 **private** 属性时，动态类通常会返回值 **undefined**，而不是生成错误。下表说明了只有在严格模式下使用点运算符访问 **private** 属性时才会生成错误：

	严格模式	标准模式
点运算符 ()	编译时错误	undefined
中括号运算符 ([])	undefined	undefined

protected 属性 (**attribute**) 是 ActionScript 3.0 中的新增属性 (**attribute**)，可使属性 (**property**) 对所属类或子类中的调用方可见。换句话说，**protected** 属性在所属类中可用，或者对继承层次结构中该类下面的类可用。无论子类在同一包中还是在不同包中，这一点都适用。

如果熟悉 ActionScript 2.0，就会了解此功能类似于 ActionScript 2.0 中的 **private** 属性。ActionScript 3.0 **protected** 属性也类似于 Java 中的 **protected** 属性，不同之处在于，Java 中的该属性还允许对同一包中的调用方进行访问。如果存在子类需要的变量或方法，但要对继承链外部的代码隐藏该变量或方法，此时 **protected** 属性会很有用。

internal 属性 (**attribute**) 是 ActionScript 3.0 的新增属性 (**attribute**)，可使属性 (**property**) 对所在包中的调用方可见。该属性是包中代码的默认属性 (**attribute**)，它适用于没有以下任意属性 (**attribute**) 的任何属性 (**property**)：

- **public**
- **private**

- `protected`
- 用户定义的命名空间

`internal` 属性与 Java 中的默认访问控制相似，不过，在 Java 中该访问级别没有明确的名称，只能通过省略所有其它访问修饰符来实现。ActionScript 3.0 中提供的 `internal` 属性 (attribute) 旨在为您提供一个明确表达自己意图的选项，以使属性 (property) 仅对所在包中的调用方可见。

static 属性

`static` 属性 (attribute) 可以与用 `var`、`const` 或 `function` 关键字声明的那些属性 (property) 一起使用，使用该属性 (attribute) 可将属性 (property) 附加到类而不是类的实例。类外部的代码必须使用类名 (而不是使用实例名) 调用静态属性 (property)。

静态属性不由子类继承，但是这些属性是子类作用域链中的一部分。这意味着在子类体中，不必引用在其中定义静态变量或方法的类，就可以使用静态变量或方法。有关详细信息，请参阅第 94 页的“[不继承静态属性](#)”。

用户定义的命名空间属性

作为预定义访问控制属性的替代方法，您可以创建自定义命名空间以用作属性。每个定义只能使用一个命名空间属性，而且不能将命名空间属性与任何访问控制属性 (`public`、`private`、`protected` 和 `internal`) 组合使用。有关使用命名空间的详细信息，请参阅第 35 页的“[命名空间](#)”。

变量

可以使用 `var` 或 `const` 关键字声明变量。在脚本的整个执行过程中，使用 `var` 关键字声明的变量可多次更改其变量值。使用 `const` 关键字声明的变量称为常量，只能赋值一次。尝试给已初始化的常量分配新值，将生成错误。有关详细信息，请参阅第 55 页的“[常数](#)”。

静态变量

静态变量是使用 `static` 关键字和 `var` 或 `const` 语句共同声明的。静态变量附加到类而不是类的实例，对于存储和共享应用于对象的整个类的信息非常有用。例如，当要保存类实例化的总次数或者要存储允许的最大类实例数，使用静态变量比较合适。

下面的示例创建一个 `totalCount` 变量 (用于跟踪类实例化数) 和一个 `MAX_NUM` 常量 (用于存储最大实例化数)。

`totalCount` 和 `MAX_NUM` 这两个变量是静态变量，因为它们包含的值应用于整个类，而不是某个特定实例。

```
class StaticVars
{
    public static var totalCount:int = 0;
    public static const MAX_NUM:uint = 16;
}
```

`StaticVars` 类及其任何子类外部的代码只能通过该类本身来引用 `totalCount` 和 `MAX_NUM` 属性。例如，以下代码有效：

```
trace(StaticVars.totalCount); // output: 0
trace(StaticVars.MAX_NUM); // output: 16
```

不能通过类实例访问静态变量，因此以下代码会返回错误：

```
var myStaticVars:StaticVars = new StaticVars();
trace(myStaticVars.totalCount); // error
trace(myStaticVars.MAX_NUM); // error
```

必须在声明常量的同时初始化使用 `static` 和 `const` 关键字声明的变量，就像 `StaticVars` 类初始化 `MAX_NUM` 那样。您不能为构造函数或实例方法中的 `MAX_NUM` 赋值。以下代码会生成错误，因为它不是初始化静态常量的有效方法：

```
// !! Error to initialize static constant this way
class StaticVars2
{
    public static const UNIQUESORT:uint;
    function initializeStatic():void
    {
        UNIQUESORT = 16;
    }
}
```

实例变量

实例变量包括使用 `var` 和 `const` 关键字但未使用 `static` 关键字声明的属性。实例变量附加到类实例而不是整个类，对于存储特定于实例的值很有用。例如，`Array` 类有一个名为 `length` 的实例属性，用来存储 `Array` 类的特定实例保存的数组元素的个数。

不能覆盖子类中声明为 `var` 或 `const` 的实例变量。但是，通过覆盖 `getter` 和 `setter` 方法，可以实现类似于覆盖变量的功能。有关详细信息，请参阅第 83 页的“[get 和 set 存取器方法](#)”。

方法

方法是类定义中的函数。创建类的一个实例后，该实例就会捆绑一个方法。与在类外部声明的函数不同，不能将方法与附加方法的实例分开使用。

方法是使用 `function` 关键字定义的。与任何类属性一样，可以向方法应用任何类属性特性，包括 `private`、`protected`、`public`、`internal`、`static` 或自定义命名空间。您可以使用如下所示的函数语句：

```
public function sampleFunction():String {}
```

或者，也可以使用分配了函数表达式的变量，如下所示：

```
public var sampleFunction:Function = function () {}
```

多数情况下，您需要使用函数语句而不是函数表达式，原因如下：

- 函数语句更为简洁易读。
- 函数语句允许使用 `override` 和 `final` 关键字。有关详细信息，请参阅第 92 页的“[覆盖方法](#)”。
- 函数语句在标识符（即函数名）与方法体代码之间创建了更强的绑定。由于可以使用赋值语句更改变量值，可随时断开变量与其函数表达式之间的连接。虽然可通过使用 `const`（不是 `var`）声明变量来解决这个问题，但这种方法并不是最好的做法，因为这会使代码难以阅读，还会禁止使用 `override` 和 `final` 关键字。

必须使用函数表达式的一种情况是：选择将函数附加到原型对象时。有关详细信息，请参阅第 99 页的“[原型对象](#)”。

构造函数方法

构造函数方法有时简单称为构造函数，是与在其中定义函数的类共享同一名称的函数。只要使用 `new` 关键字创建了类实例，就会执行构造函数方法中包括的所有代码。例如，以下代码定义名为 `Example` 的简单类，该类包含名为 `status` 的属性。`status` 变量的初始值是在构造函数中设置的。

```
class Example
{
    public var status:String;
    public function Example()
    {
        status = "initialized";
    }
}

var myExample:Example = new Example();
trace(myExample.status); // output: initialized
```

构造函数方法只能是公共方法，但可以选择性地使用 `public` 属性。不能对构造函数使用任何其它访问控制说明符（包括使用 `private`、`protected` 或 `internal`）。也不能对函数构造方法使用用户定义的命名空间。

构造函数可以使用 `super()` 语句显式地调用其直接超类的构造函数。如果未显式调用超类构造函数，编译器会在构造函数体中的第一个语句前自动插入一个调用。还可以使用 `super` 前缀作为对超类的引用来调用超类的方法。如果决定在同一构造函数中使用 `super()` 和 `super`，务必先调用 `super()`。否则，`super` 引用的行为将会与预期不符。另外，`super()` 构造函数也应在 `throw` 或 `return` 语句之前调用。

下面的示例说明如果在调用 `super()` 构造函数之前尝试使用 `super` 引用，将会发生什么情况。新类 `ExampleEx` 扩展了 `Example` 类。`ExampleEx` 构造函数尝试访问在其超类中定义的状态变量，但访问是在调用 `super()` 之前进行的。`ExampleEx` 构造函数中的 `trace()` 语句生成了 `null` 值，原因是 `status` 变量在 `super()` 构造函数执行之前不可用。

```
class ExampleEx extends Example
{
    public function ExampleEx()
    {
        trace(super.status);
        super();
    }
}

var mySample:ExampleEx = new ExampleEx(); // output: null
```

虽然在构造函数中使用 `return` 语句是合法的，但是不允许返回值。换句话说，`return` 语句不得有相关的表达式或值。因此，不允许构造函数方法返回值，这意味着不可以指定任何返回值。

如果没有在类中定义构造函数方法，编译器将会为您自动创建一个空构造函数。如果某个类扩展了另一个类，编译器将会在所生成的构造函数中包括 `super()` 调用。

静态方法

静态方法也叫做类方法，它们是使用 `static` 关键字声明的方法。静态方法附加到类而不是类的实例，因此在封装对单个实例的状态以外的内容有影响的功能时，静态方法很有用。由于静态方法附加到整个类，所以只能通过类访问静态方法，而不能通过类实例访问。

静态方法为封装所提供的功能不仅仅在影响类实例状态的方面。换句话说，如果方法提供的功能对类实例的值没有直接的影响，该方法应是静态方法。例如，`Date` 类具有名为 `parse()` 的静态方法，它接收字符串并将其转换为数字。该方法就是静态方法，因为它并不影响类的单个实例。而 `parse()` 方法使用表示日期值的字符串，分析该字符串，然后使用与 `Date` 对象的内部表示形式兼容的格式返回一个数字。此方法不是实例方法，因为将该方法应用到 `Date` 类的实例并没有任何意义。

请将静态 `parse()` 方法与 `Date` 类的一个实例方法（如 `getMonth()`）相比较。`getMonth()` 方法是一个实例方法，因为它通过检索 `Date` 实例的特定组件（即 `month`），对实例值直接执行操作。

由于静态方法不绑定到单个实例，因此不能在静态方法体中使用关键字 `this` 或 `super`。`this` 和 `super` 这两个引用只在实例方法上下文中有意义。

与其它基于类的编程语言不同，`ActionScript 3.0` 中的静态方法不是继承的。有关详细信息，请参阅第 94 页的“[不继承静态属性](#)”。

实例方法

实例方法指的是不使用 `static` 关键字声明的方法。实例方法附加到类实例而不是整个类，在实现对类的各个实例有影响的功能时，实例方法很有用。例如，`Array` 类包含名为 `sort()` 的实例方法，该实例方法直接对 `Array` 实例执行操作。

在实例方法体中，静态变量和实例变量都在作用域中，这表示使用一个简单的标识符可以引用同一类中定义的变量。例如，以下类 `CustomArray` 扩展了 `Array` 类。`CustomArray` 类定义一个名为 `arrayCountTotal` 的静态变量（用于跟踪类实例总数）、一个名为 `arrayNumber` 实例变量（用于跟踪创建实例的顺序）和一个名为 `getPosition()` 的实例方法（用于返回这两个变量的值）。

```

public class CustomArray extends Array
{
    public static var arrayCountTotal:int = 0;
    public var arrayNumber:int;

    public function CustomArray()
    {
        arrayNumber = ++arrayCountTotal;
    }

    public function getArrayPosition():String
    {
        return ("Array " + arrayNumber + " of " + arrayCountTotal);
    }
}

```

虽然类外部的代码必须使用 `CustomArray.arrayCountTotal` 通过类对象来引用 `arrayCountTotal` 静态变量，但是位于 `getPosition()` 方法体中的代码可以直接引用静态 `arrayCountTotal` 变量。即使对于超类中的静态变量，这一点也适用。虽然在 ActionScript 3.0 中不继承静态属性，但是超类的静态属性在作用域中。例如，`Array` 类有几个静态变量，其中一个是名为 `DESCENDING` 的常量。位于 `Array` 子类中的代码可以使用一个简单的标识符来引用静态常量 `DESCENDING`。

```

public class CustomArray extends Array
{
    public function testStatic():void
    {
        trace(DESCENDING); // output: 2
    }
}

```

实例方法体中的 `this` 引用的值是对方法所附加实例的引用。下面的代码说明 `this` 引用指向包含方法的实例：

```

class ThisTest
{
    function thisValue():ThisTest
    {
        return this;
    }
}

var myTest:ThisTest = new ThisTest();
trace(myTest.thisValue() == myTest); // output: true

```

使用关键字 `override` 和 `final` 可以控制实例方法的继承。可以使用 `override` 属性重新定义继承的方法，以及使用 `final` 属性禁止子类覆盖方法。有关详细信息，请参阅第 92 页的“[覆盖方法](#)”。

get 和 set 存取器方法

`get` 和 `set` 存取器函数还分别称为 `getter` 和 `setter`，可以使用这些函数为创建的类提供易于使用的编程接口，并遵循信息隐藏和封装的编程原则。使用 `get` 和 `set` 函数可保持类的私有类属性，但允许类用户访问这些属性，就像他们在访问类变量而不是调用类方法。

这种方法的好处是，可避免出现具有不实用名称的传统存取器函数，如 `getProperty()` 和 `setProperty()`。`getter` 和 `setter` 的另一个好处是，使用它们可避免允许进行读写访问的每个属性有两个面向公共的函数。

下面的示例类名为 `GetSet`，其中包含名为 `publicAccess()` 的 `get` 和 `set` 存取器函数，用于提供对名为 `privateProperty` 的私有变量的访问：

```

class GetSet
{
    private var privateProperty:String;

    public function get publicAccess():String
    {
        return privateProperty;
    }

    public function set publicAccess(setValue:String):void
    {
        privateProperty = setValue;
    }
}

```

如果尝试直接访问属性 `privateProperty`, 将会发生错误, 如下所示:

```

var myGetSet:GetSet = new GetSet();
trace(myGetSet.privateProperty); // error occurs

```

`GetSet` 类的用户所使用的对象显示为名为 `publicAccess` 的属性, 但实际上这是对名为 `privateProperty` 的 `private` 属性执行的一对 `get` 和 `set` 存取器函数。下面的示例将实例化 `GetSet` 类, 然后使用名为 `publicAccess` 的公共存取器设置 `privateProperty` 的值:

```

var myGetSet:GetSet = new GetSet();
trace(myGetSet.publicAccess); // output: null
myGetSet.publicAccess = "hello";
trace(myGetSet.publicAccess); // output: hello

```

使用 `getter` 和 `setter` 函数还可以覆盖从超类继承来的属性, 这是使用常规类成员变量时不能做到的。在子类中不能覆盖使用 `var` 关键字声明的类成员变量。但是, 使用 `getter` 和 `setter` 函数创建的属性没有此限制。可以对从超类继承的 `getter` 和 `setter` 函数使用 `override` 属性。

绑定方法

绑定方法有时也叫做闭包方法, 就是从它的实例提取的方法。作为参数传递给函数的方法或作为值从函数返回的方法都是绑定方法。在 ActionScript 3.0 中, 新增的绑定方法类似于闭包函数, 其中保留了词汇环境, 即使从其实例中提取出来也是如此。绑定方法与闭包函数之间的主要不同差别是, 绑定函数的 `this` 引用保留到实现方法的实例的链接或绑定。换句话说, 绑定方法中的 `this` 引用总是指向实现方法的原始对象。对于闭包函数, `this` 引用是通用的, 这意味着调用函数时, 该引用指向与函数关联的任何对象。

如果使用 `this` 关键字, 了解绑定方法就很重要。重新调用 `this` 关键字可提供对方法父对象的引用。大多数 ActionScript 程序员都希望 `this` 关键字总是引用包含方法定义的对象或类。但是, 如果不使用方法绑定, 并不是总是做到这样。例如, 在以前版本的 ActionScript 中, `this` 引用并不总是引用实现方法的实例。从 ActionScript 2.0 的实例中提取方法后, 不但 `this` 引用不绑定到原始实例, 而且实例类的成员变量和方法也不可用。在 ActionScript 3.0 中不存在这样的问题, 这是因为将方法当作参数传递时会自动创建绑定方法。绑定方法用于确保 `this` 关键字总是引用在其中定义了方法的对象或类。

下面的代码定义了名为 `ThisTest` 的类, 该类包含一个名为 `foo()` 的方法 (该方法定义绑定方法) 和一个名为 `bar()` 的方法 (该方法返回绑定方法)。类外部的代码创建 `ThisTest` 类的实例, 然后调用 `bar()` 方法, 最后将返回值存储在名为 `myFunc` 的变量中。

```

class ThisTest
{
    private var num:Number = 3;
    function foo():void // bound method defined
    {
        trace("foo's this: " + this);
        trace("num: " + num);
    }
    function bar():Function
    {
        return foo; // bound method returned
    }
}

var myTest:ThisTest = new ThisTest();
var myFunc:Function = myTest.bar();
trace(this); // output: [object global]
myFunc();
/* output:
foo's this: [object ThisTest]
output: num: 3 */

```

代码的最后两行表明：虽然前一行中的 `this` 引用指向全局对象，但绑定方法 `foo()` 中的 `this` 引用仍然指向 `ThisTest` 类的实例。另外，存储在 `myFunc` 变量中的绑定方法仍然可以访问 `ThisTest` 类的成员变量。如果以上代码在 ActionScript 2.0 中运行，`this` 引用会匹配，但 `num` 变量将为 `undefined`。

绑定方法最值得注意的一种情况是使用事件处理函数，因为 `addEventListener()` 方法要求将函数或方法作为参数来传递。有关详细信息，请参阅第 214 页的“[事件侦听器](#)”中定义为类方法的 `Listener` 函数。

类的枚举

“枚举”是您创建的一些自定义数据类型，用于封装一小组值。ActionScript 3.0 并不支持具体的枚举工具，这与 C++ 使用 `enum` 关键字或 Java 使用 `Enumeration` 接口不一样。不过，您可以使用类或静态常量创建枚举。例如，ActionScript 3.0 中的 `PrintJob` 类使用名为 `PrintJobOrientation` 的枚举来存储由 “landscape” 和 “portrait” 组成的一组值，如下面的代码所示：

```

public final class PrintJobOrientation
{
    public static const LANDSCAPE:String = "landscape";
    public static const PORTRAIT:String = "portrait";
}

```

按照惯例，枚举类是使用 `final` 属性声明的，因为不需要扩展该类。该类仅由静态成员组成，这表示不创建该类的实例。而是直接通过类对象来访问枚举值，如以下代码摘录中所示：

```

var pj:PrintJob = new PrintJob();
if(pj.start())
{
    if (pj.orientation == PrintJobOrientation.PORTRAIT)
    {
        ...
    }
    ...
}

```

ActionScript 3.0 中的所有枚举类都只包含 `String`、`int` 或 `uint` 类型的变量。使用枚举而不使用文本字符串或数字值的好处是，使用枚举更易于发现字面错误。如果枚举名输入错误，ActionScript 编译器会生成一个错误。如果使用字面值，存在拼写错误或使用了错误数字时，编译器并不会报错。在上一个示例中，如果枚举常量的名称不正确，编译器会生成错误，如以下代码摘录中所示：

```
if (pj.orientation == PrintJobOrientation.PORTRAI) // compiler error
```

但是，如果拼错了字符串字面值，编译器并不生成错误，如下所示：

```
if (pj.orientation == "portrai") // no compiler error
```

创建枚举的第二种方法还包括使用枚举的静态属性创建单独的类。这种方法的不同之处在于每一个静态属性都包含一个类实例，而不是字符串或整数值。例如，以下代码为一星期中的各天创建了一个枚举类：

```
public final class Day
{
    public static const MONDAY:Day = new Day();
    public static const TUESDAY:Day = new Day();
    public static const WEDNESDAY:Day = new Day();
    public static const THURSDAY:Day = new Day();
    public static const FRIDAY:Day = new Day();
    public static const SATURDAY:Day = new Day();
    public static const SUNDAY:Day = new Day();
}
```

ActionScript 3.0 并不使用这种方法，但是许多开发人员都使用，他们更喜欢使用这种方法提供的改进类型检查功能。例如，返回枚举值的方法可将返回值限定为枚举数据类型。以下代码不但显示了返回星期中各天的函数，还显示了将枚举类型用作类型注释的函数调用：

```
function getDay():Day
{
    var date:Date = new Date();
    var retDay:Day;
    switch (date.day)
    {
        case 0:
            retDay = Day.MONDAY;
            break;
        case 1:
            retDay = Day.TUESDAY;
            break;
        case 2:
            retDay = Day.WEDNESDAY;
            break;
        case 3:
            retDay = Day.THURSDAY;
            break;
        case 4:
            retDay = Day.FRIDAY;
            break;
        case 5:
            retDay = Day.SATURDAY;
            break;
        case 6:
            retDay = Day.SUNDAY;
            break;
    }
    return retDay;
}

var dayOfWeek:Day = getDay();
```

您还可以增强 Day 类的功能，以使其将一个整数与星期中的各天关联，并提供一个 `toString()` 方法来返回各天的字符串表示形式。您可能希望实践一下，采用这种方法来增强 Day 类的功能。

嵌入资源类

ActionScript 3.0 使用称为嵌入资源类的特殊类来表示嵌入的资源。嵌入资源指的编译时包括在 SWF 文件中的资源，如声音、图像或字体。嵌入资源而不是动态加载资源，可以确保资源在运行时可用，但代价是增加了 SWF 文件的大小。

在 Flash 中使用嵌入资源类

要嵌入资源，首先将该资源放入 FLA 文件的库中。接着，使用资源的链接属性，提供资源的嵌入资源类的名称。如果无法在类路径中找到具有该名称的类，则将自动生成一个类。然后，可以创建嵌入资源类的实例，并使用任何由该类定义或继承的属性和方法。例如，以下代码可用于播放链接到名为 PianoMusic 的嵌入资源类的嵌入声音：

```
var piano:PianoMusic = new PianoMusic();
var sndChannel:SoundChannel = piano.play();
```

接口

接口是方法声明的集合，以使不相关的对象能够彼此通信。例如，ActionScript 3.0 定义了 IEventDispatcher 接口，其中包含的方法声明可供类用于处理事件对象。IEventDispatcher 接口建立了标准方法，供对象相互传递事件对象。以下代码显示 IEventDispatcher 接口的定义：

```
public interface IEventDispatcher
{
    function addEventListener(type:String, listener:Function,
        useCapture:Boolean=false, priority:int=0,
        useWeakReference:Boolean = false):void;
    function removeEventListener(type:String, listener:Function,
        useCapture:Boolean=false):void;
    function dispatchEvent(event:Event):Boolean;
    function hasEventListener(type:String):Boolean;
    function willTrigger(type:String):Boolean;
}
```

接口的基础是方法的接口与方法的实现之间的区别。方法的接口包括调用该方法必需的所有信息，包括方法名、所有参数和返回类型。方法的实现不仅包括接口信息，而且还包括执行方法的行为的可执行语句。接口定义只包含方法接口，实现接口的所有类负责定义方法实现。

在 ActionScript 3.0 中，EventDispatcher 类通过定义所有 IeventDispatcher 接口方法并在每个方法中添加方法体来实现 IEventDispatcher 接口。以下代码摘录自 EventDispatcher 类定义：

```
public class EventDispatcher implements IEventDispatcher
{
    function dispatchEvent(event:Event):Boolean
    {
        /* implementation statements */
    }

    ...
}
```

IEventDispatcher 接口用作一个协议，EventDispatcher 实例通过该协议处理事件对象，然后将事件对象传递到也实现了 IeventDispatcher 接口的其它对象。

另一种描述接口的方法是：接口定义了数据类型，就像类一样。因此，接口可以用作类型注释，也像类一样。作为数据类型，接口还可以与需要指定数据类型的运算符一起使用，如 **is** 和 **as** 运算符。但是与类不同的是，接口不可以实例化。这个区别使很多程序员认为接口是抽象的数据类型，认为类是具体的数据类型。

定义接口

接口定义的结构类似于类定义的结构，只是接口只能包含方法但不能包含方法体。接口不能包含变量或常量，但是可以包含 **getter** 和 **setter**。要定义接口，请使用 **interface** 关键字。例如，接口 IExternalizable 是 ActionScript 3.0 中 flash.utils 包的一部分。IExternalizable 接口定义一个用于对对象进行序列化（即将对象转换为适合在设备上存储或通过网络传输的格式）的协议。

```
public interface IExternalizable
{
    function writeExternal(output:IDataOutput):void;
    function readExternal(input:IDataInput):void;
}
```

注意，**IExternalizable** 接口是使用 **public** 访问控制修饰符声明的。只能使用 **public** 和 **internal** 访问控制说明符来修饰接口定义。接口定义中的方法声明不能包含任何访问控制说明符。

ActionScript 3.0 遵循一种约定，其中接口名以大写字母 I 开头，不过您可以使用任何合法的标识符作为接口名。接口定义经常位于包的顶级。接口定义不能放在类定义或另一个接口定义中。

接口可扩展一个或多个其它接口。例如，下面的接口 **IExample** 扩展了 **IExternalizable** 接口：

```
public interface IExample extends IExternalizable
{
    function extra():void;
}
```

实现 **IExample** 接口的所有类不但必须包括 **extra()** 方法的实现，还要包括从 **IExternalizable** 接口继承的 **writeExternal()** 和 **readExternal()** 方法的实现。

在类中实现接口

类是唯一可实现接口的 ActionScript 3.0 语言元素。在类声明中使用 **implements** 关键字可实现一个或多个接口。下面的示例定义两个接口 **IAlpha** 和 **IBeta** 以及实现这两个接口的类 **Alpha**：

```
interface IAlpha
{
    function foo(str:String):String;
}

interface IBeta
{
    function bar():void;
}

class Alpha implements IAlpha, IBeta
{
    public function foo(param:String):String {}
    public function bar():void {}
}
```

在实现接口的类中，实现的方法必须：

- 使用 **public** 访问控制标识符。
- 使用与接口方法相同的名称。
- 拥有相同数量的参数，每一个参数的数据类型都要与接口方法参数的数据类型相匹配。
- 使用相同的返回类型。

```
public function foo(param:String):String {}
```

不过在命名所实现方法的参数时，您有一定的灵活性。虽然实现的方法的参数数和每个参数的数据类型必须与接口方法的参数数和数据类型相匹配，但参数名不需要匹配。例如，在上一个示例中，将 **Alpha.foo()** 方法的参数命名为 **param**：

但是，将 **IAlpha.foo()** 接口方法中的参数命名为 **str**：

```
function foo(str:String):String;
```

另外，使用默认参数值也具有一定的灵活性。接口定义可以包含使用默认参数值的函数声明。实现这种函数声明的方法必须采用默认参数值，默认参数值是与接口定义中指定的值具有相同数据类型的一个成员，但是实际值不一定匹配。例如，以下代码定义的接口包含一个使用默认参数值 3 的方法：

```
interface IGamma
{
    function doSomething(param:int = 3):void;
}
```

以下类定义实现 `IGamma` 接口，但使用不同的默认参数值：

```
class Gamma implements IGamma
{
    public function doSomething(param:int = 4):void {}
}
```

提供这种灵活性的原因是，实现接口的规则的设计目的是确保数据类型兼容性，因此不必要求采用相同的参数名和默认参数名，就能实现目标。

继承

继承是指一种代码重用的形式，允许程序员基于现有类开发新类。现有类通常称为基类或超类，新类通常称为子类。继承的主要优势是，允许重复使用基类中的代码，但不修改现有代码。此外，继承不要求改变其它类与基类交互的方式。不必修改可能已经过彻底测试或可能已被使用的现有类，使用继承可将该类视为一个集成模块，可使用其它属性或方法对它进行扩展。因此，您使用 `extends` 关键字指明类从另一类继承。

通过继承还可以在代码中利用多态。有一种方法在应用于不同数据类型时会有不同行为，多态就是对这样的方法应用一个方法名的能力。名为 `Shape` 的基类就是一个简单的示例，该类有名为 `Circle` 和 `Square` 的两个子类。`Shape` 类定义了名为 `area()` 的方法，该方法返回形状的面积。如果已实现多态，则可以对 `Circle` 和 `Square` 类型的对象调用 `area()` 方法，然后执行正确的计算。使用继承能实现多态，实现的方式是允许子类继承和重新定义或覆盖基类中的方法。在下面的示例中，由 `Circle` 和 `Square` 两个类重新定义了 `area()` 方法：

```

class Shape
{
    public function area():Number
    {
        return NaN;
    }
}

class Circle extends Shape
{
    private var radius:Number = 1;
    override public function area():Number
    {
        return (Math.PI * (radius * radius));
    }
}

class Square extends Shape
{
    private var side:Number = 1;
    override public function area():Number
    {
        return (side * side);
    }
}

var cir:Circle = new Circle();
trace(cir.area()); // output: 3.141592653589793
var sq:Square = new Square();
trace(sq.area()); // output: 1

```

因为每个类定义一个数据类型，所以使用继承会在基类和扩展基类的类之间创建一个特殊关系。子类保证拥有其基类的所有属性，这意味着子类的实例总是可以替换基类的实例。例如，如果方法定义了 Shape 类型的参数 (parameter)，由于 Circle 扩展了 Shape，因此 Circle 类型的参数 (argument) 是合法的，如下所示：

```

function draw(shapeToDraw:Shape) {}

var myCircle:Circle = new Circle();
draw(myCircle);

```

实例属性和继承

对于实例属性 (property)，无论是使用 function、var 还是使用 const 关键字定义的，只要在基类中未使用 private 属性 (attribute) 声明该属性 (property)，这些属性都可以由子类继承。例如，ActionScript 3.0 中的 Event 类具有许多子类，它们继承了所有事件对象共有的属性。

对于某些类型的事件，Event 类包含了定义事件所需的所有属性。这些类型的事件不需要 Event 类中定义的实例属性以外的实例属性。complete 事件和 connect 事件就是这样的事件，前者在成功加载数据时发生，后者在建立网络连接时发生。

下面的示例是从 Event 类中摘录的，显示由子类继承的某些属性和方法。由于继承了属性，因此任何子类的实例都可以访问这些属性。

```

public class Event
{
    public function get type():String;
    public function get bubbles():Boolean;
    ...

    public function stopPropagation():void {}
    public function stopImmediatePropagation():void {}
    public function preventDefault():void {}
    public function isDefaultPrevented():Boolean {}

    ...
}

```

其它类型的事件需要 `Event` 类中没有提供的特有属性。这些事件是使用 `Event` 类的子类定义的，所以可向 `Event` 类中定义的属性添加新属性。`MouseEvent` 类就是这样的一个子类，它可添加与鼠标移动或鼠标单击相关的事件的特有属性，如 `mouseMove` 和 `click` 事件。下面的示例是从 `MouseEvent` 类中摘录的，它说明了在子类中存在但在基类中不存在的属性的定义：

```

public class MouseEvent extends Event
{
    public static const CLICK:String= "click";
    public static const MOUSE_MOVE:String = "mouseMove";
    ...

    public function get stageX():Number {}
    public function get stageY():Number {}
    ...
}

```

访问控制说明符和继承

如果某一属性是用 `public` 关键字声明的，则该属性对任何位置的代码可见。这表示 `public` 关键字与 `private`、`protected` 和 `internal` 关键字不同，它对属性继承没有任何限制。

如果属性是使用 `private` 关键字声明的，该属性只在定义该属性的类中可见，这表示它不能由任何子类继承。此行为与以前版本的 ActionScript 不同，在这些版本中，`private` 关键字的行为更类似于 ActionScript 3.0 中的 `protected` 关键字。

`protected` 关键字指出某一属性不仅在定义该属性的类中可见，而且还在所有子类中可见。与 Java 编程语言中的 `protected` 关键字不一样，ActionScript 3.0 中的 `protected` 关键字并不使属性对同一包中的所有其它类可见。在 ActionScript 3.0 中，只有子类可以访问使用 `protected` 关键字声明的属性。此外，`protected` 属性对子类可见，不管子类和基类是在同一包中，还是在不同包中。

要限制某一属性在定义该属性的包中的可见性，请使用 `internal` 关键字或者不使用任何访问控制说明符。未指定访问控制说明符时，应用的默认访问控制说明符是 `internal` 访问控制说明符。标记为 `internal` 的属性将只由位于在同一包中的子类继承。

可以使用下面的示例来查看每一个访问控制说明符如何影响跨越包边界的继承。以下代码定义了一个名为 `AccessControl` 的主应用程序类和两个名为 `Base` 的 Extender 的其它类。`Base` 类在名为 `foo` 的包中，`Extender` 类是 `Base` 类的子类，它在名为 `bar` 的包中。`AccessControl` 类只导入 `Extender` 类，并创建一个 `Extender` 实例，该实例试图访问在 `Base` 类中定义的名为 `str` 的变量。`str` 变量将声明为 `public` 以便代码能够进行编译和运行，如以下摘录中所示：

```

// Base.as in a folder named foo
package foo
{
    public class Base
    {
        public var str:String = "hello"; // change public on this line
    }
}

// Extender.as in a folder named bar
package bar
{
    import foo.Base;
    public class Extender extends Base
    {
        public function getString():String {
            return str;
        }
    }
}

// main application class in file named AccessControl.as
package
{
    import flash.display.MovieClip;
    import bar.Extender;
    public class AccessControl extends MovieClip
    {
        public function AccessControl()
        {
            var myExt:Extender = new Extender();
            trace(myExt.str); // error if str is not public
            trace(myExt.getString()); // error if str is private or internal
        }
    }
}

```

要查看其它访问控制说明符如何影响先前示例的编译和执行，请在 AccessControl 类中删除或注释掉以下行，然后将 str 变量的访问控制说明符更改为 `private`、`protected` 或 `internal`：

```
trace(myExt.str); // error if str is not public
```

不允许覆盖变量

将继承使用 `var` 或 `const` 关键字声明的属性，但不能对其进行覆盖。覆盖某一属性就表示在子类中重新定义该属性。唯一可覆盖的属性类型是方法，即使用 `function` 关键字声明的属性。虽然不能覆盖实例变量，但是通过为实例变量创建 `getter` 和 `setter` 方法并覆盖这些方法，可实现类似的功能。有关详细信息，请参阅第 92 页的“[覆盖方法](#)”。

覆盖方法

覆盖方法表示重新定义已继承方法的行为。静态方法不能继承，也不能覆盖。但是，实例方法可由子类继承，也可覆盖，只要符合以下两个条件：

- 实例方法在基类中不是使用 `final` 关键字声明的。当 `final` 关键字与实例方法一起使用时，该关键字指明程序员的设计目的是要禁止子类覆盖方法。
- 实例方法在基类中不是使用 `private` 访问控制说明符声明的。如果某个方法在基类中标记为 `private`，则在子类中定义同名方法时不需要使用 `override` 关键字，因为基类方法在子类中不可见。

要覆盖符合这些条件的实例方法，子类中的方法定义必须使用 `override` 关键字，且必须在以下几个方面与方法的超类版本相匹配：

- 覆盖方法必须与基类方法具有相同级别的访问控制。标记为内部的方法与没有访问控制说明符的方法具有相同级别的访问控制。
- 覆盖方法必须与基类方法具有相同的参数数。
- 覆盖方法参数必须与基类方法参数具有相同的数据类型注释。
- 覆盖方法必须与基类方法具有相同的返回类型。

但是，覆盖方法中的参数名不必与基类中的参数名相匹配，只要参数数和每个参数的数据类相匹配即可。

super 语句

覆盖方法时，程序员经常希望在要覆盖的超类方法的行为上添加行为，而不是完全替换该行为。这需要通过某种机制来允许子类中的方法调用它本身的超类版本。**super** 语句就提供了这样一种机制，其中包含对直接超类的引用。下面的示例定义了名为 **Base** 的类（其中包含名为 **thanks()** 的方法），还包含名为 **Extender** 的 **Base** 类的子类（用于覆盖 **thanks()** 方法）。**Extender.thanks()** 方法使用 **super** 语句调用 **Base.thanks()**。

```
package {
    import flash.display.MovieClip;
    public class SuperExample extends MovieClip
    {
        public function SuperExample()
        {
            var myExt:Extender = new Extender();
            trace(myExt.thanks()); // output: Mahalo nui loa
        }
    }
}

class Base {
    public function thanks():String
    {
        return "Mahalo";
    }
}

class Extender extends Base
{
    override public function thanks():String
    {
        return super.thanks() + " nui loa";
    }
}
```

覆盖 **getter** 和 **setter**

虽然不能覆盖超类中定义的变量，但是可以覆盖 **getter** 和 **setter**。例如，下面的代码用于覆盖在 ActionScript 3.0 的 **MovieClip** 类中定义的名为 **currentLabel** 的 **getter**：

```

package
{
    import flash.display.MovieClip;
    public class OverrideExample extends MovieClip
    {
        public function OverrideExample()
        {
            trace(currentLabel)
        }
        override public function get currentLabel():String
        {
            var str:String = "Override: ";
            str += super.currentLabel;
            return str;
        }
    }
}

```

OverrideExample 类构造函数中的 trace() 语句的输出是 Override: null，这说明该示例能够覆盖继承的 currentLabel 属性。

不继承静态属性

静态属性不由子类继承。这意味着不能通过子类的实例访问静态属性。只能通过定义静态属性的类对象来访问静态属性。例如，以下代码定义了名为 Base 的基类和扩展名为 Extender 的 Base 子类。Base 类中定义了名为 test 的静态变量。以下摘录中编写的代码在严格模式下不会进行编译，在标准模式下会生成运行时错误。

```

package {
    import flash.display.MovieClip;
    public class StaticExample extends MovieClip
    {
        public function StaticExample()
        {
            var myExt:Extender = new Extender();
            trace(myExt.test); // error
        }
    }
}

class Base {
    public static var test:String = "static";
}

class Extender extends Base { }

```

访问静态变量 test 的唯一方式是通过类对象，如以下代码所示：

```
Base.test;
```

但允许使用与静态属性相同的名称定义实例属性。可以在与静态属性相同的类中或在子类中定义这样的实例属性。例如，以上示例中的 Base 类可以具有一个名为 test 的实例属性。在以下代码中，由于 Extender 类继承了实例属性，因此代码能够进行编译和执行。如果 test 实例变量的定义移到（不是复制） Extender 类，该代码也会编译并执行。

```

package
{
    import flash.display.MovieClip;
    public class StaticExample extends MovieClip
    {
        public function StaticExample()
        {
            var myExt:Extender = new Extender();
            trace(myExt.test); // output: instance
        }
    }
}

class Base
{
    public static var test:String = "static";
    public var test:String = "instance";
}

class Extender extends Base {}

```

静态属性和作用域链

虽然并不继承静态属性，但是静态属性在定义它们的类或该类的任何子类的作用域链中。同样，可以认为静态属性在定义它们的类和任何子类的作用域中。这意味着在定义静态属性的类体及该类的任何子类中可直接访问静态属性。

下面的示例对上一个示例中定义的类进行了修改，以说明 `Base` 类中定义的 `test` 静态变量在 `Extender` 类的作用域中。换句话说，`Extender` 类可以访问 `test` 静态变量，而不必用定义 `test` 的类名作为变量的前缀。

```

package
{
    import flash.display.MovieClip;
    public class StaticExample extends MovieClip
    {
        public function StaticExample()
        {
            var myExt:Extender = new Extender();
        }
    }
}

class Base {
    public static var test:String = "static";
}

class Extender extends Base
{
    public function Extender()
    {
        trace(test); // output: static
    }
}

```

如果使用与同类或超类中的静态属性相同的名称定义实例属性，则实例属性在作用域链中的优先级比较高。因此认为实例属性遮蔽了静态属性，这意味着会使用实例属性的值，而不使用静态属性的值。例如，以下代码显示如果 `Extender` 类定义名为 `test` 的实例变量，`trace()` 语句将使用实例变量的值，而不使用静态变量的值：

```

package
{
    import flash.display.MovieClip;
    public class StaticExample extends MovieClip
    {
        public function StaticExample()
        {
            var myExt:Extender = new Extender();
        }
    }
}

class Base
{
    public static var test:String = "static";
}

class Extender extends Base
{
    public var test:String = "instance";
    public function Extender()
    {
        trace(test); // output: instance
    }
}

```

高级主题

本节开始先简单介绍 ActionScript 和 OOP 的历史，然后讨论 ActionScript 3.0 对象模型，以及该模型如何启用新的 ActionScript 虚拟机 (AVM2) 显著提供运行速度（与包含旧 ActionScript 虚拟机 (AVM1) 的以前版本的 Flash Player 相比）。

ActionScript OOP 支持的历史

由于 ActionScript 3.0 是在以前版本的 ActionScript 基础上构建的，了解 ActionScript 对象模型的发展过程可能有所帮助。ActionScript 最初作为早期版本的 Flash 创作工具的简单编写脚本机制。后来，程序员开始使用 ActionScript 建立更加复杂的应用程序。为了迎合这些程序员的需要，每个后续版本都添加了一些语言功能以帮助创建复杂的应用程序。

ActionScript 1.0

ActionScript 1.0 指在 Flash Player 6 和更早版本中使用的语言版本。即使在这个早期开发阶段，ActionScript 对象模型也是建立在基础数据类型对象的概念的基础上。ActionScript 对象是由一组属性 构成的复合数据类型。讨论对象模型时，术语属性 包括附加到对象的所有内容，如变量、函数或方法。

尽管第一代 ActionScript 不支持使用 `class` 关键字定义类，但是可以使用称为原型对象的特殊对象来定义类。Java 和 C++ 等基于类的语言中使用 `class` 关键字创建要实例化为具体对象的抽象类定义，而 ActionScript 1.0 等基于原型的语言则将现有对象用作其它对象的模型（或原型）。基于类的语言中的对象可能指向作为其模板的类，而基于原型的语言中的对象则指向作为其模板的另一个对象（即其原型）。

要在 ActionScript 1.0 中创建类，可以为该类定义一个构造函数。在 ActionScript 中，函数不只是抽象定义，还是实际对象。您创建的构造函数用作该类实例的原型对象。以下代码创建一个名为 `Shape` 的类，还定义一个名为 `visible` 的属性，该属性默认情况下设置为 `true`：

```
// base class
function Shape() {}
// Create a property named visible.
Shape.prototype.visible = true;
```

此构造函数定义可以使用 new 运算符实例化的 Shape 类，如下所示：

```
myShape = new Shape();
```

就像 Shape() 构造函数对象用作 Shape 类实例的原型一样，它还可以用作 Shape 的子类（即扩展 Shape 类的其它类）的原型。


创建作为 Shape 类的子类的类的过程分两个步骤。首先，通过定义类的构造函数创建该类，如下所示：

```
// child class
function Circle(id, radius)
{
    this.id = id;
    this.radius = radius;
}
```

然后，使用 new 运算符将 Shape 类声明为 Circle 类的原型。默认情况下，创建的所有类都使用 Object 类作为其原型，这意味着 Circle.prototype 当前包含一个通用对象（Object 类的实例）。要指定 Circle 的原型是 Shape 而不是 Object，请使用以下代码更改 Circle.prototype 的值，使其包含 Shape 对象而不是通用对象。

```
// Make Circle a subclass of Shape.
Circle.prototype = new Shape();
```

Shape 类和 Circle 类现在通过通常所说的原型链的继承关系联系在一起。下图说明了原型链中的关系：



每个原型链末端的基类是 Object 类。Object 类包含一个名为 Object.prototype 的静态属性，该属性指向在 ActionScript 1.0 中创建的所有对象的基原型对象。我们的示例原型链中的下一个对象是 Shape 对象。这是因为从不显式设置 Shape.prototype 属性，所以它仍然包含通用对象（Object 类的实例）。此链中的最后一个链环是 Circle 类，该类链接到其原型 Shape 类（Circle.prototype 属性包含 Shape 对象）。

如果创建了 Circle 类的实例（如下面的示例所示），该实例会继承 Circle 类的原型链：

```
// Create an instance of the Circle class.
myCircle = new Circle();
```

回想一下，我们创建了一个名为 visible 的属性作为 Shape 类的成员。在示例中，visible 属性并不作为 myCircle 对象的一部分，只是 Shape 对象的一个成员，但以下代码行的输出为 true：

```
trace(myCircle.visible); // output: true
```

Flash Player 能够沿着原型链检查 myCircle 对象是否继承了 visible 属性。执行此代码时，Flash Player 首先在 myCircle 对象的属性中搜索名为 visible 的属性，但是未发现这样的属性。Flash Player 然后在下一个 Circle.prototype 对象中查找，但是仍未发现名为 visible 的属性。继续检查原型链，Flash Player 最终发现了在 Shape.prototype 对象上定义的 visible 属性，并输出该属性的值。

为了简便，本节省略了原型链的很多难懂之处和细节，目的是为了提供足够的信息帮助您了解 ActionScript 3.0 对象模型。

ActionScript 2.0

在 ActionScript 2.0 中引入了 class、extends、public 和 private 等新关键字，通过使用这些关键字，您可以按 Java 和 C++ 等基于类的语言用户所熟悉的方式来定义类。ActionScript 1.0 和 ActionScript 2.0 中的基础继承机制是相同的，了解这一点很重要。ActionScript 2.0 只是增加了定义类的新语法。在该语言的两个版本中，原型链的作用方式是一样的。

ActionScript 2.0 中引入了新语法（如以下摘录中所示），可允许以数程序员认为更直观的方式定义类：

```
// base class
class Shape
{
    var visible:Boolean = true;
}
```

注意，ActionScript 2.0 还引入了用于编译时类型检查的类型注释。使用类型注释，可以将上个示例中的 visible 属性声明为应只包含布尔值。新 extends 关键字还简化了创建子类的过程。在下面的示例中，通过使用 extends 关键字，可以一步完成在 ActionScript 1.0 中需要分两步完成的过程：

```
// child class
class Circle extends Shape
{
    var id:Number;
    var radius:Number;
    function Circle(id, radius)
    {
        this.id = id;
        this.radius = radius;
    }
}
```


构造函数现在声明为类定义的一部分，还必须显式声明类属性 id 和 radius。

ActionScript 2.0 中还增加了对接口定义的支持，使您能够使用为对象间通信正式定义的协议来进一步改进面向对象的程序。

ActionScript 3.0 类对象

常见的面向对象的编程范例多数常与 Java 和 C++ 相关联，这种范例使用类定义对象的类型。采用这种范例的编程语言也趋向使用类来构造类定义的数据类型的实例。ActionScript 使用类是为了实现以上两个目的，但其根本还是一种基于原型的语言，并带有有趣的特征。ActionScript 为每一类定义创建了特殊的类对象，允许共享行为和状态。但是，对多数 ActionScript 程序员而言，这个特点可能与实际编码没有什么牵连。ActionScript 3.0 的设计目的是，不使用（甚至是不必了解）这些特殊类对象，就可创建复杂的面向对象的 ActionScript 应用程序。对于要利用类对象的高级程序员，本节提供有关问题的深入讨论。

下图显示一个类对象的结构，该类对象表示使用语句 class A {} 定义的名为 A 的简单类：



图中的每个矩形表示一个对象。图中的每一个对象都有下标字符 A，这表示该对象属于类 A。类对象 (CA) 包含对许多其它重要对象的引用。实例 traits 对象 (TA) 用于存储在类定义中定义的实例属性。类 traits 对象 (TCA) 表示类的内部类型，用于存储该类定义的静态属性（下标字符 C 代表“类”）。原型对象 (PA) 始终指的是最初通过 constructor 属性附加到的类对象。

traits 对象

traits 对象是 ActionScript 3.0 中的新增对象，它是为了提高性能而实现的。在以前版本的 ActionScript 中，名称查找是一个耗时的过程，因为 Flash Player 要搜索原型链。在 ActionScript 3.0 中，名称查找更有效、耗时更少，因为可以将继承属性从超类复制到子类的 traits 对象。

编程代码不能直接访问 traits 对象，但是性能和内存使用情况的改善可反映它的存在。traits 对象给 AVM2 提供了关于类的布局和内容的详细信息。借助这些信息，AVM2 可显著减少执行时间，因为它可以经常生成直接机器指令来直接访问属性或直接调用方法，而省去了查找名称所耗费的时间。

由于使用了 traits 对象，与以前版本中 ActionScript 类似对象相比，该版本中对象占用内存的时间明显减少。例如，如果某个类已密封（即，该类未声明为 dynamic），则该类实例不需要动态添加属性的哈希表，只保留一个到 traits 对象的指针和该类中定义的固定属性的某些位置。因此，如果对象在 ActionScript 2.0 中需要占用 100 个字节的内存，在 ActionScript 3.0 中只需要占用 20 个字节的内存。

注：traits 对象是内部实现详细信息，不保证在将来版本的 ActionScript 中此对象不更改，甚至消失。

原型对象

每个 ActionScript 类对象都有一个名为 prototype 属性，它表示对类的原型对象的引用。ActionScript 根本上是基于原型的语言，原型对象是旧内容。有关详细信息，请参阅 第 96 页的“[ActionScript OOP 支持的历史](#)”。

prototype 属性是只读属性，这表示不能将其修改为指向其它对象。这不同于以前版本 ActionScript 中的类 prototype 属性，在以前版本中可以重新分配 prototype，使它指向其它类。虽然 prototype 属性是只读属性，但是它所引用的原型对象不是只读的。换句话说，可以向原型对象添加新属性。向原型对象添加的属性可在类的所有实例中共享。

原型链是以前版本的 ActionScript 的唯一继承机制，在 ActionScript 3.0 中，原型链只具有辅助作用。主要的继承机制（固定属性）是由 traits 对象在内部处理的。固定属性指的是定义为类定义的一部分的变量或方法。固定属性继承也叫做类继承，因为它是与 class、extends 和 override 等关键字相关的继承机制。

原型链提供了另一种继承机制，该机制的动态性比固定属性继承的更强。既可以将属性作为类定义的一部分，也可以在运行时通过类对象的 prototype 属性向类的原型对象中添加属性。但是，请注意，如果将编译器设置为严格模式，则不能访问添加到原型对象中的属性，除非使用 dynamic 关键字声明类。

Object 类就是这样类的示例，它的原型对象附加了若干属性。Object 类的 `toString()` 和 `valueOf()` 方法实际上是一些函数，它们分配给 Object 类原型对象的属性。以下是一个示例，说明这些方法的声明理论上是怎样的（实际实现时会因实现详细信息而稍有不同）：

```
public dynamic class Object
{
    prototype.toString = function()
    {
        // statements
    };
    prototype.valueOf = function()
    {
        // statements
    };
}
```

如前所述，可以将属性附加到类定义外部的类原型对象。例如，也可以在 Object 类定义外部定义 `toString()` 方法，如下所示：

```
Object.prototype.toString = function()
{
    // statements
};
```

但是，原型继承与固定属性继承不一样，如果要重新定义子类中的方法，原型继承不需要 `override` 关键字。例如。如果要重新定义 `Object` 类的子类中的 `valueOf()` 方法，您有以下三种选择。第一，可以在类定义中的子类原型对象上定义 `valueOf()` 方法。以下代码创建一个名为 `Foo` 的 `Object` 子类，并将 `Foo` 原型对象的 `valueOf()` 方法重新定义为类定义的一部分。因为每个类都是从 `Object` 继承的，所以不需要使用 `extends` 关键字。

```
dynamic class Foo
{
    prototype.valueOf = function()
    {
        return "Instance of Foo";
    };
}
```

第二，可以在类定义外部对 `Foo` 原型对象定义 `valueOf()` 方法，如以下代码中所示：

```
Foo.prototype.valueOf = function()
{
    return "Instance of Foo";
};
```

第三，可以将名为 `valueOf()` 的固定属性定义为 `Foo` 类的一部分。这种方法与其它混合了固定属性继承与原型继承的方法有所不同。要重新定义 `valueOf()` 的 `Foo` 的任何子类必须使用 `override` 关键字。以下代码显示 `valueOf()` 定义为 `Foo` 中的固定属性：

```
class Foo
{
    function valueOf():String
    {
        return "Instance of Foo";
    }
}
```

AS3 命名空间

由于存在两种继承机制，即固定属性继承和原型继承，所以涉及到核心类的属性和方法时，就存在两种机制的兼容性问题。如果与 ActionScript 所基于的 ECMAScript 语言规范兼容，则要求使用原型继承，这意味着核心类的属性和方法是在该类的原型对象上定义的。另一方面，如果与 ActionScript 3.0 兼容，则要求使用固定属性继承，这意味着核心类的属性和方法是使用 `const`、`var` 和 `function` 关键字在类定义中定义的。此外，如果使用固定属性而不是原型属性，将显著提升运行时性能。

在 ActionScript 3.0 中，通过同时将原型继承和固定属性继承用于核心类，解决了这个问题。每一个核心类都包含两组属性和方法。一组是在原型对象上定义的，用于与 ECMAScript 规范兼容，另一组使用固定属性定义和 AS3 命名空间定义，以便与 ActionScript 3.0 兼容。

AS3 命名空间提供了一种约定机制，用来在两组属性和方法之间做出选择。如果不使用 AS3 命名空间，核心类的实例会继承在核心类的原型对象上定义的属性和方法。如果决定使用 AS3 命名空间，核心类的实例会继承 AS3 版本，因为固定属性的优先级始终高于原型属性。换句话说，只要固定属性可用，则始终使用固定属性，而不使用同名的原型属性。

通过用 AS3 命名空间限定属性或方法，可以选择使用 AS3 命名空间版本的属性或方法。例如，下面的代码使用 AS3 版本的 `Array.pop()` 方法：

```
var nums:Array = new Array(1, 2, 3);
nums.AS3::pop();
trace(nums); // output: 1,2
```

或者，也可以使用 `use namespace` 指令打开代码块中所有定义的 AS3 命名空间。例如，以下代码使用 `use namespace` 指令为 `pop()` 和 `push()` 方法打开 AS3 命名空间：

```
use namespace AS3;

var nums:Array = new Array(1, 2, 3);
nums.pop();
nums.push(5);
trace(nums) // output: 1,2,5
```

ActionScript 3.0 还为每组属性提供了编译器选项，以便将 AS3 命名空间应用于整个程序。-as3 编译器选项表示 AS3 命名空间，-es 编译器选项表示原型继承选项（es 代表 ECMAScript）。要打开整个程序的 AS3 命名空间，请将 -as3 编译器选项设置为 true，将 -es 编译器选项设置为 false。要使用原型版本，请将编译器选项设置为相反值。Adobe Flex Builder 3 和 Adobe Flash CS4 Professional 的默认编译器选项是 -as3 = true 和 -es = false。

如果计划扩展任何核心类并覆盖任何方法，应了解 AS3 命名空间对声明覆盖方法的方式有什么影响。如果要使用 AS3 命名空间，覆盖核心类方法的任何方法都必须使用 AS3 命名空间以及 override 属性。如果不打算使用 AS3 命名空间且要重新定义子类中的核心类方法，则不应使用 AS3 命名空间或 override 关键字。

示例：GeometricShapes

GeometricShapes 范例应用程序说明了如何使用 ActionScript 3.0 来应用很多面向对象的概念和功能，其中包括：

- 定义类
- 扩展类
- 多态和 override 关键字
- 定义、扩展和实现接口

示例中还包括一个用于创建类实例的“工厂方法”，说明如何将返回值声明为接口的实例，以及通过一般方法使用返回的对象。

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。在 Samples/GeometricShapes 文件夹下可找到 GeometricShapes 应用程序文件。该应用程序包含以下文件：

文件	描述
GeometricShapes.mxml 或 GeometricShapes.fla	Flash 或 Flex 中的主应用程序文件（分别为 FLA 和 MXML）。
com/example/programmingas3/geometricshapes/IGeometricShape.as	由所有 GeometricShapes 应用程序类实现的基接口定义方法。
com/example/programmingas3/geometricshapes/IPolygon.as	由有多条边的 GeometricShapes 应用程序类实现的接口定义方法。
com/example/programmingas3/geometricshapes/RegularPolygon.as	一种几何形状，这种几何形状的边长相等，并且这些边围绕形状中心对称分布。
com/example/programmingas3/geometricshapes/Circle.as	一种用于定义圆的几何形状。
com/example/programmingas3/geometricshapes/EquilateralTriangle.as	RegularPolygon 的子类，用于定义所有边长相等的三角形。
com/example/programmingas3/geometricshapes/Square.as	RegularPolygon 的子类，用于定义所有四条边相等的矩形。
com/example/programmingas3/geometricshapes/GeometricShapeFactory.as	包含工厂方法的一个类，用于创建给定了形状类型和尺寸的形状。


定义 GeometricShapes 类

使用 GeometricShapes 应用程序可以允许用户指定几何形状的类型和尺寸。随后，应用程序通过形状描述、其面积和周长进行响应。

应用程序的用户界面显得很零碎，其中包括一些可用于选择形状类型、设置尺寸及显示描述的控件。这个应用程序最引人注意的部分是程序的内部，即在类和接口本身的结构中。

这个应用程序用于处理几何形状，但是它并不以图形化方式显示这些形状。它提供一个由类和接口组成的小型库，后面的章节会重复使用这个库（请参阅第 259 页的“示例：SpriteArranger”）。SpriteArranger 示例以图形化方式显示形状，还允许用户根据 GeometricShapes 应用程序中提供的类框架处理形状。

在下图中使用统一建模语言 (UML) 表示法显示此示例中用于定义几何形状的类和接口：



GeometricShapes 示例类

定义接口的通用行为

这个 GeometricShapes 应用程序处理三种形状：圆、正方形和等边三角形。GeometricShapes 类结构以非常简单的接口 IGeometricShape 作为开始，它列出通用于所有这三种形状的方法。

```

package com.example.programmingas3.geometricshapes
{
    public interface IGeometricShape
    {
        function getArea():Number;
        function describe():String;
    }
}
  
```

该接口定义两个方法：getArea() 方法和 describe() 方法，前者计算并返回形状的面积，后者汇编形状属性的文本描述。

我们还要知道每个形状的周边长度。但是，圆的周边长度叫做周长，它的计算方式是独有的，所以其行为异于三角形或正方形的行为。不过，三角形、正方形和其它多边形之间仍然有很多类似之处，所以为它们定义一个新接口类 IPolygon 还是很有意义的。IPolygon 接口也相当简单，如下所示：

```
package com.example.programmingas3.geometricshapes
{
    public interface IPolygon extends IGeometricShape
    {
        function getPerimeter():Number;
        function getSumOfAngles():Number;
    }
}
```

这个接口定义了所有多边形的两个通用方法：`getPerimeter()` 方法和 `getSumOfAngles()` 方法，前者用于计算所有边相加后的总长，后者用于将所有内角相加起来。

`IPolygon` 接口扩展了 `IGeometricShape` 接口，这意味着实现 `IPolygon` 接口的所有类必须声明所有四个方法，即来自 `IGeometricShape` 接口的两个方法和来自 `IPolygon` 接口的两个方法。

定义形状类

对每种形状的方法有所了解后，就可以定义形状类本身了。就需要实现的方法数而言，最简单的形状是 `Circle` 类，如下所示：

```
package com.example.programmingas3.geometricshapes
{
    public class Circle implements IGeometricShape
    {
        public var diameter:Number;

        public function Circle(diam:Number = 100):void
        {
            this.diameter = diam;
        }

        public function getArea():Number
        {
            // The formula is Pi * radius * radius.
            var radius:Number = diameter / 2;
            return Math.PI * radius * radius;
        }

        public function getCircumference():Number
        {
            // The formula is Pi * diameter.
            return Math.PI * diameter;
        }

        public function describe():String
        {
            var desc:String = "This shape is a Circle.\n";
            desc += "Its diameter is " + diameter + " pixels.\n";
            desc += "Its area is " + getArea() + ".\n";
            desc += "Its circumference is " + getCircumference() + ".\n";
            return desc;
        }
    }
}
```

`Circle` 类用于实现 `IGeometricShape` 接口，因此，它必须为 `getArea()` 方法和 `describe()` 方法提供代码。此外，它还定义 `getCircumference()` 方法，该方法对 `Circle` 类是唯一的。另外，`Circle` 类还声明属性 `diameter`，其它多边形类没有该属性。

其它两种形状，即正方形和等边三角形的某些其它方面是共同的：它们各自的边长相等，各自都有可用于计算周长和内角总和的通用公式。实际上，这些通用公式还适用于将来需要定义的任何其它正多边形。

`RegularPolygon` 类将是 `Square` 和 `EquilateralTriangle` 两个类的超类。使用超类可在同一个位置定义通用方法，所以不必在每个子类中单独定义方法。以下是 `RegularPolygon` 类的代码：

```

package com.example.programmingas3.geometricshapes
{
    public class RegularPolygon implements IPolygon
    {
        public var numSides:int;
        public var sideLength:Number;

        public function RegularPolygon(len:Number = 100, sides:int = 3):void
        {
            this.sideLength = len;
            this.numSides = sides;
        }

        public function getArea():Number
        {
            // This method should be overridden in subclasses.
            return 0;
        }

        public function getPerimeter():Number
        {
            return sideLength * numSides;
        }

        public function getSumOfAngles():Number
        {
            if (numSides >= 3)
            {
                return ((numSides - 2) * 180);
            }
            else
            {
                return 0;
            }
        }

        public function describe():String
        {
            var desc:String = "Each side is " + sideLength + " pixels long.\n";
            desc += "Its area is " + getArea() + " pixels square.\n";
            desc += "Its perimeter is " + getPerimeter() + " pixels long.\n";
            desc += "The sum of all interior angles in this shape is " + getSumOfAngles() + " degrees.\n";
            return desc;
        }
    }
}

```

第一，RegularPolygon 类用于声明所有正多边形的两个通用属性：每边的长度（sideLength 属性）和边数（numSides 属性）。

RegularPolygon 类实现 IPolygon 接口，还声明 IPolygon 接口的所有四个方法。它使用通用公式来实现 getPerimeter() 和 getSumOfAngles() 两种方法。

因为用于 getArea() 方法的公式将因形状的不同而有所不同，所以该方法的基本版本不能包括可由子类方法继承的通用逻辑。而是只返回 0 默认值来指明未计算面积。要正确地计算每一个形状的面积，RegularPolygon 类的子类本身一定要覆盖 getArea() 方法本身。

EquilateralTriangle 类的以下代码说明如何覆盖 getArea() 方法：

```

package com.example.programmingas3.geometricshapes
{
    public class EquilateralTriangle extends RegularPolygon
    {
        public function EquilateralTriangle(len:Number = 100):void
        {
            super(len, 3);
        }

        public override function getArea():Number
        {
            // The formula is ((sideLength squared) * (square root of 3)) / 4.
            return ( (this.sideLength * this.sideLength) * Math.sqrt(3) ) / 4;
        }

        public override function describe():String
        {
            /* starts with the name of the shape, then delegates the rest
            of the description work to the RegularPolygon superclass */
            var desc:String = "This shape is an equilateral Triangle.\n";
            desc += super.describe();
            return desc;
        }
    }
}

```

`override` 关键字指示 `EquilateralTriangle.getArea()` 方法有意覆盖 `RegularPolygon` 超类中的 `getArea()` 方法。当调用 `EquilateralTriangle.getArea()` 方法时，该方法使用先前代码中的公式计算面积，从不执行 `RegularPolygon.getArea()` 方法中的代码。

不同的是，`EquilateralTriangle` 类并不定义它自己版本的 `getPerimeter()` 方法。当调用 `EquilateralTriangle.getPerimeter()` 方法时，调用会沿继承链找到 `RegularPolygon` 超类的 `getPerimeter()` 方法，并执行其中的代码。

`EquilateralTriangle()` 构造函数使用 `super()` 语句显式地调用其超类的 `RegularPolygon()` 构造函数。如果这两个构造函数使用同一组参数，则可以完全省略 `EquilateralTriangle()` 构造函数，而执行 `RegularPolygon()` 构造函数。但是，`RegularPolygon()` 构造函数需要额外的参数，即 `numSides`。所以，`EquilateralTriangle()` 函数调用 `super(len, 3)`，该语句传递 `len` 输入参数和值 3，以指示三角形有 3 个边。

`describe()` 方法也使用 `super()` 语句，但是采用不同的方式调用 `describe()` 方法的 `RegularPolygon` 超类版本。`EquilateralTriangle.describe()` 方法先将 `desc` 字符串变量设置为有关形状类型的语句。然后调用 `super.describe()` 来获取 `RegularPolygon.describe()` 方法的结果，之后将结果追加到 `desc` 字符串。

此处不详细讨论 `Square` 类，但它类似于 `EquilateralTriangle` 类，它们都提供构造函数及其自己的 `getArea()` 和 `describe()` 方法实现。

多态和工厂方法

可以采用许多有趣的方法正确使用接口的一组类及继承。例如，到目前为止讨论的所有形状类不是用于实现 `IGeometricShape` 接口，就是用于扩展执行的超类。因此，如果将一个变量定义为 `IGeometricShape` 实例，您不必知道它实际是 `Circle` 类的实例还是 `Square` 类的实例，就可以调用它的 `describe()` 方法。

以下代码说明这是如何实现的：

```

var myShape:IGeometricShape = new Circle(100);
trace(myShape.describe());

```

当调用 `myShape.describe()` 时，它执行 `Circle.describe()` 方法，因为即使将变量定义为 `IGeometricShape` 接口的实例，`Circle` 也是它的基础类。

这个示例说明了运行时多态的原则：完全相同的方法调用会导致执行不同的代码，这取决于要调用了其方法的对象的类。

GeometricShapes 应用程序会使用设计模式的简化版本（叫做“工厂方法”）应用这种基于接口的多态原则。术语工厂方法指的是一个函数，该函数返回一个对象，其基本数据类型或内容可能会因上下文而不同。

以下显示的 **GeometricShapeFactory** 类定义一个名为 `createShape()` 的工厂方法：

```
package com.example.programmingas3.geometricshapes
{
    public class GeometricShapeFactory
    {
        public static var currentShape:IGeometricShape;

        public static function createShape(shapeName:String,
                                         len:Number):IGeometricShape
        {
            switch (shapeName)
            {
                case "Triangle":
                    return new EquilateralTriangle(len);

                case "Square":
                    return new Square(len);

                case "Circle":
                    return new Circle(len);
            }
            return null;
        }

        public static function describeShape(shapeType:String, shapeSize:Number):String
        {
            GeometricShapeFactory.currentShape =
                GeometricShapeFactory.createShape(shapeType, shapeSize);
            return GeometricShapeFactory.currentShape.describe();
        }
    }
}
```

`createShape()` 工厂方法允许形状子类构造函数定义所创建的实例的详细信息，同时返回新对象作为 **IGeometricShape** 实例，以便应用程序可以采用更通用的方式处理这些对象。

前面示例中的 `describeShape()` 方法说明应用程序如何使用工厂方法来获取对更具体对象的一般引用。该应用程序可通过以下方式获取新创建的 **Circle** 对象的描述：

```
GeometricShapeFactory.describeShape("Circle", 100);
```

然后，`describeShape()` 方法使用相同的参数调用 `createShape()` 工厂方法，以将新的 **Circle** 对象存储在名为 `currentShape` 的静态变量（该变量的类型是 **IGeometricShape** 对象）中。接下来，对 `currentShape` 调用 `describe()` 方法，该调用将被自动解析以执行 `Circle.describe()` 方法，从而返回圆的详细描述。

增强范例应用程序

增强或更改应用程序后，接口和继承的实际作用会非常明显。

假定要在这个示例应用程序中添加新形状，即一个五边形。您需要创建一个新 **Pentagon** 类，以扩展 **RegularPolygon** 类并定义它自己版本的 `getArea()` 和 `describe()` 方法。然后，在应用程序用户界面的组合框中添加一个新 **Pentagon** 选项。就是这样，**Pentagon** 类将通过继承来自动获取 **RegularPolygon** 类的 `getPerimeter()` 方法和 `getSumOfAngles()` 方法的功能。由于该类是

从实现 `IGeometricShape` 接口的类继承的，因此 `Pentagon` 实例也可以视为 `IGeometricShape` 实例。这意味着，要添加新形状类型，不需要更改 `GeometricShapeFactory` 类中的任何方法的方法签名（因而也不需要更改使用 `GeometricShapeFactory` 类的任何代码）。

您可能希望实践一下，将 `Pentagon` 类添加到 `Geometric Shapes` 示例，以了解向应用程序中添加新功能时，使用接口和继承将如何减轻工作量。

第 6 章：处理日期和时间

时间可能并不代表一切，但它在软件应用程序中通常是一个关键要素。ActionScript 3.0 提供了多种强大的手段来管理日历日期、时间和时间间隔。以下两个主类提供了大部分的计时功能：`Date` 类和 `flash.utils` 包中的新 `Timer` 类。

日期和时间基础知识

处理日期和时间简介

日期和时间是在 ActionScript 程序中使用的一种常见信息类型。例如，您可能需要了解当前星期值，或测量用户在特定屏幕上花费多少时间，并且还可能会执行很多其它操作。在 ActionScript 中，可以使用 `Date` 类来表示某一时刻，其中包含日期和时间信息。`Date` 实例中包含各个日期和时间单位的值，其中包括年、月、日、星期、小时、分钟、秒、毫秒以及时区。对于更高级的用法，ActionScript 还包括 `Timer` 类，您可以使用该类在一定延迟后执行动作，或按重复间隔执行动作。

常见日期和时间任务

本章介绍了以下常见的日期和时间信息处理任务：

- 处理 `Date` 对象
- 获取当前日期和时间
- 访问各个日期和时间单位（日、年、小时、分钟等）
- 使用日期和时间执行运算
- 在时区之间进行转换
- 执行重复动作
- 在设定的时间间隔后执行动作

重要概念和术语

以下参考列表包含将会在本章中遇到的重要术语：

- UTC 时间：通用协调时间，即“零小时”基准时区。所有其它时区均被定义为相对于 UTC 时间快或慢一定的小时数。

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。由于本章中的代码清单主要用于处理 `Date` 对象，测试示例将涉及查看示例中使用的变量值，方法是：通过将值写入舞台上的文本字段实例，或使用 `trace()` 函数将值输出到“输出”面板。第 29 页的“[测试本章内的示例代码清单](#)”将详细介绍这些方法。

管理日历日期和时间

ActionScript 3.0 的所有日历日期和时间管理函数都集中在顶级 `Date` 类中。`Date` 类包含一些方法和属性，这些方法和属性能够使您按照通用协调时间 (UTC) 或特定于时区的本地时间来处理日期和时间。UTC 是一种标准时间定义，它实质上与格林尼治标准时间 (GMT) 相同。

创建 Date 对象

Date 类是所有核心类中构造函数方法形式最为多变的类之一。您可以用以下四种方式来调用 Date 类。

第一，如果未给定参数，则 Date() 构造函数将按照您所在时区的本地时间返回包含当前日期和时间的 Date 对象。这里提供了一个示例：

```
var now:Date = new Date();
```

第二，如果仅给定了一个数字参数，则 Date() 构造函数将其视为自 1970 年 1 月 1 日以来经过的毫秒数，并且返回对应的 Date 对象。请注意，您传入的毫秒值将被视为自 1970 年 1 月 1 日（UTC 时间）以来经过的毫秒数。但是，该 Date 对象会按照您所在的本地时区来显示值，除非您使用特定于 UTC 的方法来检索和显示这些值。如果仅使用一个毫秒参数来创建新的 Date 对象，则应确保考虑到您的当地时间与 UTC 之间的时区差异。以下语句创建一个设置为 1970 年 1 月 1 日午夜（UTC 时间）的 Date 对象：

```
var millisecondsPerDay:int = 1000 * 60 * 60 * 24;  
// gets a Date one day after the start date of 1/1/1970  
var startTime:Date = new Date(millisecondsPerDay);
```

第三，您可以将多个数值参数传递给 Date() 构造函数。该构造函数将这些参数分别视为年、月、日、小时、分钟、秒和毫秒，并将返回一个对应的 Date 对象。假定这些输入参数采用的是本地时间而不是 UTC。以下语句获取一个设置为 2000 年 1 月 1 日开始的午夜（本地时间）的 Date 对象：

```
var millennium:Date = new Date(2000, 0, 1, 0, 0, 0, 0);
```

第四，您可以将单个字符串参数传递给 Date() 构造函数。该构造函数将尝试把字符串解析为日期或时间部分，然后返回对应的 Date 对象。如果使用此方法，最好将 Date() 构造函数包含在 try..catch 块中，以捕获所有分析错误。Date() 构造函数接受多种不同的字符串格式，如《ActionScript 3.0 语言和组件参考》中所列。以下语句使用字符串值初始化一个新的 Date 对象：

```
var nextDay:Date = new Date("Mon May 1 2006 11:30:00 AM");
```

如果 Date() 构造函数无法成功分析该字符串参数，它不会引发异常。但是，所得到的 Date 对象将包含一个无效的日期值。

获取时间单位值

可以使用 Date 类的属性或方法从 Date 对象中提取各种时间单位的值。下面的每个属性为您提供了 Date 对象中的一个时间单位的值：

- fullYear 属性
- month 属性，以数字格式表示，分别以 0 到 11 表示一月到十二月
- date 属性，表示月中某一天的日历数字，范围为 1 到 31
- day 属性，以数字格式表示一周中的某一天，其中 0 表示星期日
- hours 属性，范围为 0 到 23
- minutes 属性
- seconds 属性
- milliseconds 属性

实际上，Date 类为您提供了解获取这些值的多种方式。例如，您可以用四种不同方式获取 Date 对象的月份值：

- month 属性
- getMonth() 方法
- monthUTC 属性
- getMonthUTC() 方法

所有四种方式实质上具有同等的效率，因此您可以任意使用一种最适合应用程序的方法。

刚才列出的属性表示总日期值的各个部分。例如，`milliseconds` 属性永远不会大于 999，因为当它达到 1000 时，秒钟值就会增加 1 并且 `milliseconds` 属性会重置为 0。

如果要获取 `Date` 对象自 1970 年 1 月 1 日 (UTC) 起所经过毫秒数的值，可以使用 `getTime()` 方法。通过使用与其相对应的 `setTime()` 方法，可以使用自 1970 年 1 月 1 日 (UTC) 起所经过的毫秒数更改现有 `Date` 对象的值。

执行日期和时间运算

您可以使用 `Date` 类对日期和时间执行加法和减法运算。日期值在内部以毫秒的形式保存，因此您应将其它值转换成毫秒，然后再将它们与 `Date` 对象进行加减。

如果应用程序将执行大量的日期和时间运算，您可能会发现创建常量来保存常见时间单位值（以毫秒的形式）非常有用，如下所示：

```
public static const millisecondsPerMinute:int = 1000 * 60;
public static const millisecondsPerHour:int = 1000 * 60 * 60;
public static const millisecondsPerDay:int = 1000 * 60 * 60 * 24;
```

现在，可以方便地使用标准时间单位来执行日期运算。下列代码使用 `getTime()` 和 `setTime()` 方法将日期值设置为当前时间一个小时后的时间：

```
var oneHourFromNow:Date = new Date();
oneHourFromNow.setTime(oneHourFromNow.getTime() + millisecondsPerHour);
```

设置日期值的另一种方式是仅使用一个毫秒参数创建新的 `Date` 对象。例如，下列代码将一个日期加上 30 天以计算另一个日期：

```
// sets the invoice date to today's date
var invoiceDate:Date = new Date();

// adds 30 days to get the due date
var dueDate:Date = new Date(invoiceDate.getTime() + (30 * millisecondsPerDay));
```

接着，将 `millisecondsPerDay` 常量乘以 30 以表示 30 天的时间，将结果与 `invoiceDate` 值相加，用于设置 `dueDate` 值。

在时区之间进行转换

在需要将日期从一种时区转换成另一种时区时，使用日期和时间运算十分方便。也可以使用 `getTimezoneOffset()` 方法，该方法返回的值表示 `Date` 对象的时区与 UTC 之间相差的分钟数。此方法之所以返回以分钟为单位的值，是因为并不是所有时区之间都正好相差一个小时，有些时区与邻近的时区仅相差半个小时。

以下示例使用时区偏移量将日期从本地时间转换成 UTC。该示例首先以毫秒为单位计算时区值，然后按照该量调整 `Date` 值：

```
// creates a Date in local time
var nextDay:Date = new Date("Mon May 1 2006 11:30:00 AM");

// converts the Date to UTC by adding or subtracting the time zone offset
var offsetMilliseconds:Number = nextDay.getTimezoneOffset() * 60 * 1000;
nextDay.setTime(nextDay.getTime() + offsetMilliseconds);
```

控制时间间隔

使用 Adobe Flash CS4 Professional 开发应用程序时，您可以访问时间轴，这会使您稳定且逐帧地完成该应用程序。但在纯 ActionScript 项目中，您必须依靠其它计时机制。

循环与计时器之比较

在某些编程语言中，必须使用循环语句（如 `for` 或 `do..while`）来设计自己的计时方案。`while`。

通常，循环语句会以本地计算机所允许的速度尽可能快地执行，这表明应用程序在某些计算机上的运行速度较快而在其它计算机上则较慢。如果应用程序需要一致的计时间隔，则您需要将其与实际的日历或时钟时间联系在一起。许多应用程序（如游戏、动画和实时控制器）需要在不同计算机上均能保持一致的、规则的时间驱动计时机制。

ActionScript 3.0 的 `Timer` 类提供了一个功能强大的解决方案。使用 ActionScript 3.0 事件模型，`Timer` 类在每次达到指定的时间间隔时都会调度计时器事件。

Timer 类

在 ActionScript 3.0 中处理计时函数的首选方式是使用 `Timer` 类 (`flash.utils.Timer`)，可以使用它在每次达到间隔时调度事件。

要启动计时器，请先创建 `Timer` 类的实例，并告诉它每隔多长时间生成一次计时器事件以及在停止前生成多少次事件。

例如，下列代码创建一个每秒调度一个事件且持续 60 秒的 `Timer` 实例：

```
var oneMinuteTimer:Timer = new Timer(1000, 60);
```

`Timer` 对象在每次达到指定的间隔时都会调度 `TimerEvent` 对象。`TimerEvent` 对象的事件类型是 `timer`（由常量 `TimerEvent.TIMER` 定义）。`TimerEvent` 对象包含的属性与标准 `Event` 对象包含的属性相同。

如果将 `Timer` 实例设置为固定的间隔数，则在达到最后一次间隔时，它还会调度 `timerComplete` 事件（由常量 `TimerEvent.TIMER_COMPLETE` 定义）。

以下是一个用来展示 `Timer` 类实际操作的小示例应用程序：

```
package
{
    import flash.display.Sprite;
    import flash.events.TimerEvent;
    import flash.utils.Timer;

    public class ShortTimer extends Sprite
    {
        public function ShortTimer()
        {
            // creates a new five-second Timer
            var minuteTimer:Timer = new Timer(1000, 5);

            // designates listeners for the interval and completion events
            minuteTimer.addEventListener(TimerEvent.TIMER, onTick);
            minuteTimer.addEventListener(TimerEvent.TIMER_COMPLETE, onTimerComplete);

            // starts the timer ticking
            minuteTimer.start();
        }

        public function onTick(event:TimerEvent):void
        {
            // displays the tick count so far
            // The target of this event is the Timer instance itself.
            trace("tick " + event.target.currentCount);
        }

        public function onTimerComplete(event:TimerEvent):void
        {
            trace("Time's Up!");
        }
    }
}
```

创建 `ShortTimer` 类时，它会创建一个用于每秒计时一次并持续五秒的 `Timer` 实例。然后，它将两个侦听器添加到计时器：一个用于侦听每次计时，另一个用于侦听 `timerComplete` 事件。

接着，它启动计数器计时，并且从此时起以一秒钟的间隔执行 `onTick()` 方法。

`onTick()` 方法只显示当前的时间计数。五秒钟后，执行 `onTimerComplete()` 方法，告诉您时间已到。

运行该示例时，您应会看到下列行以每秒一行的速度显示在控制台或跟踪窗口中：

```
tick 1
tick 2
tick 3
tick 4
tick 5
Time's Up!
```

flash.utils 包中的计时函数

ActionScript 3.0 包含许多与 ActionScript 2.0 提供的计时函数类似的计时函数。这些函数是作为 `flash.utils` 包中的包级别函数提供的，它们的功能与 ActionScript 2.0 中完全相同。

函数	描述
clearInterval(id:uint):void	取消指定的 setInterval() 调用。
clearTimeout(id:uint):void	取消指定的 setTimeout() 调用。
getTimer():int	返回自 Adobe® Flash® Player 或 Adobe® AIR™ 初始化以来经过的毫秒数。
setInterval(closure:Function, delay:Number, ... arguments):uint	以指定的间隔（以毫秒为单位）运行函数。
setTimeout(closure:Function, delay:Number, ... arguments):uint	在指定的延迟（以毫秒为单位）后运行指定的函数。

这些函数仍保留在 ActionScript 3.0 以实现向后兼容。Adobe 不建议您在新的 ActionScript 3.0 应用程序中使用这些函数。通常，在应用程序中使用 Timer 类会更容易且更有效。

示例：简单的模拟时钟

简单的模拟时钟示例说明了在本章中讨论的以下两个日期和时间概念：

- 获取当前日期和时间并提取小时、分钟和秒的值
- 使用 Timer 设置应用程序的运行速度

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可以在 Samples/SimpleClock 文件夹中找到 SimpleClock 应用程序文件。该应用程序包含以下文件：

文件	描述
SimpleClockApp.mxml 或 SimpleClockApp.fla	Flash 或 Flex 中的主应用程序文件（分别为 FLA 和 MXML）。
com/example/programmingas3/simpleclock/SimpleClock.as	主应用程序文件。
com/example/programmingas3/simpleclock/AnalogClockFace.as	根据时间绘制一个圆形的钟面以及时针、分针和秒针。

定义 SimpleClock 类

此时钟示例很简单，但是将即使很简单的应用程序也组织得十分有条理是一种很好的做法，以便您将来能够很轻松地扩展这些应用程序。为此，SimpleClock 应用程序使用 SimpleClock 类处理启动和时间保持任务，然后使用另一个名称为 AnalogClockFace 的类来实际显示该时间。

以下代码用于定义和初始化 SimpleClock 类（请注意，在 Flash 版本中，SimpleClock 扩展了 Sprite 类）：

```
public class SimpleClock extends UIComponent
{
    /**
     * The time display component.
     */
    private var face:AnalogClockFace;

    /**
     * The Timer that acts like a heartbeat for the application.
     */
    private var ticker:Timer;
```

该类具有两个重要的属性：

- `face` 属性，它是 `AnalogClockFace` 类的实例
- `ticker` 属性，它是 `Timer` 类的实例

`SimpleClock` 类使用默认构造函数。`initClock()` 方法处理实际的设置工作，它创建钟面并启动 `Timer` 实例的计时。

创建钟面

`SimpleClock` 代码中后面的几行代码创建用于显示时间的钟面：

```
/**
 * Sets up a SimpleClock instance.
 */
public function initClock(faceSize:Number = 200)
{
    // creates the clock face and adds it to the display list
    face = new AnalogClockFace(Math.max(20, faceSize));
    face.init();
    addChild(face);

    // draws the initial clock display
    face.draw();
}
```

可以将钟面的大小传递给 `initClock()` 方法。如果未传递 `faceSize` 值，则使用 200 个像素的默认大小。

接着，应用程序将钟面初始化，然后使用从 `DisplayObject` 类继承的 `addChild()` 方法将该钟面添加到显示列表。然后，它调用 `AnalogClockFace.draw()` 方法显示一次钟面，同时显示当前时间。

启动计时器

创建钟面后，`initClock()` 方法会设置一个计时器：

```
// creates a Timer that fires an event once per second
ticker = new Timer(1000);

// designates the onTick() method to handle Timer events
ticker.addEventListener(TimerEvent.TIMER, onTick);

// starts the clock ticking
ticker.start();
```

首先，该方法初始化一个每秒（每隔 1000 毫秒）调度一次事件的 `Timer` 实例。由于没有向 `Timer()` 构造函数传递第二个 `repeatCount` 参数，因此 `Timer` 将无限期地重复计时。

`SimpleClock.onTick()` 方法将在每秒收到 `timer` 事件时执行一次。

```
public function onTick(event:TimerEvent):void
{
    // updates the clock display
    face.draw();
}
```

`AnalogClockFace.draw()` 方法仅绘制钟面和指针。

显示当前时间

`AnalogClockFace` 类中大多数代码都与设置钟面的显示元素有关。`AnalogClockFace` 在进行初始化时，会绘制一个圆形轮廓，将数字文本标签放在每个小时刻度处，然后创建三个 `Shape` 对象，分别表示时钟的时针、分针和秒针。

在 `SimpleClock` 应用程序运行后，它会每秒调用一次 `AnalogClockFace.draw()` 方法，如下所示：

```
/**  
 * Called by the parent container when the display is being drawn.  
 */  
public override function draw():void  
{  
    // stores the current date and time in an instance variable  
    currentTime = new Date();  
    showTime(currentTime);  
}
```

此方法将当前时间保存在变量中，因此在绘制时钟指针的过程中无法改变时间。然后，调用 `showTime()` 方法以显示指针，如下所示：

```
/**  
 * Displays the given Date/Time in that good old analog clock style.  
 */  
public function showTime(time:Date):void  
{  
    // gets the time values  
    var seconds:uint = time.getSeconds();  
    var minutes:uint = time.getMinutes();  
    var hours:uint = time.getHours();  
  
    // multiplies by 6 to get degrees  
    this.secondHand.rotation = 180 + (seconds * 6);  
    this.minuteHand.rotation = 180 + (minutes * 6);  
  
    // Multiply by 30 to get basic degrees, then  
    // add up to 29.5 degrees (59 * 0.5)  
    // to account for the minutes.  
    this.hourHand.rotation = 180 + (hours * 30) + (minutes * 0.5);  
}
```

首先，此方法提取当前时间的小时、分钟和秒的值。然后使用这些值来计算每个指针的角度。由于秒针会在 60 秒内旋转一圈，因此它每秒都会旋转 6 度 ($360/60$)。分针每分钟都旋转同样的度数。

时针每分钟都在更新，因此时针能够随着分针的跳动显示出某些时间变化过程。时针每小时旋转 30 度 ($360/12$)，但也会每分钟旋转半度（30 度除以 60 分钟）。

第 7 章：处理字符串

String 类包含使您能够使用文本字符串的方法。在处理许多对象时，字符串都十分重要。本章介绍的方法对于处理在对象（如 **TextField**、**StaticText**、**XML**、**ContextMenu** 和 **FileReference** 对象）中使用的字符串很有帮助。

字符串是字符的序列。ActionScript 3.0 支持 ASCII 字符和 Unicode 字符。

字符串基础知识

处理字符串简介

在编程语言中，字符串是指一个文本值，即串在一起而组成单个值的一系列字母、数字或其它字符。例如，以下一行代码创建一个数据类型为 **String** 的变量，并为该变量赋予一个文本字符串值：

```
var albumName:String = "Three for the money";
```

正如此示例所示，在 ActionScript 中，可使用双引号或单引号将本文引起来以表示字符串值。以下是另外几个字符串示例：

```
"Hello"  
"555-7649"  
"http://www.adobe.com/"
```

每当在 ActionScript 中处理一段文本时，您都会用到字符串值。ActionScript **String** 类是一种可用来处理文本值的数据类型。**String** 实例通常用于很多其它 ActionScript 类中的属性、方法参数等。

常见的字符串处理任务

以下是本章中讨论的与字符串有关的常见任务：

- 创建 **String** 对象
- 处理特殊字符，如回车符、制表符和非键盘字符
- 测量字符串长度
- 隔离字符串中的各个字符
- 连接字符串
- 比较字符串
- 查找、提取以及替换字符串的各部分
- 使字符串变为大写或小写

重要概念和术语

以下参考列表包含将会在本章中遇到的重要术语：

- ASCII：用于在计算机程序中表示文本字符和符号的系统。ASCII 系统支持 26 个字母英文字母表，以及有限的一组其它字符。
- 字符：文本数据的最小单位（单个字母或符号）。
- 连接：通过将一个字符串值添加到另一个字符串值结尾，将多个字符串值连接在一起，从而创建一个新的字符串值。

- 空字符串：不包含任何文本、空白或其它字符的字符串，可以写为 ""。空字符串值不同于具有 **null** 值的 **String** 变量；为 **null** 的 **String** 变量是指没有赋予 **String** 实例的变量，而空字符串则包含一个实例，其值不包含任何字符。
- 字符串：一个文本值（字符序列）。
- 字符串文本或文本字符串：在代码中显式编写的字符串值，编写为用双引号或单引号引起起来的文本值。
- 子字符串：作为另一个字符串一部分的字符串。
- **Unicode**：用于在计算机程序中表示文本字符和符号的标准系统。**Unicode** 系统允许使用任何编写系统中的任何字符。

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。由于本章中的代码清单主要用于操作文本，测试示例将涉及查看示例中使用的变量值，方法是：通过将值写入舞台上的文本字段实例，或使用 **trace()** 函数将值输出到“输出”面板。第 29 页的“[测试本章内的示例代码清单](#)”将详细介绍这些方法。

创建字符串

在 ActionScript 3.0 中，**String** 类用于表示字符串（文本）数据。ActionScript 字符串支持 ASCII 字符和 **Unicode** 字符。创建字符串的最简单方式是使用字符串文本。要声明字符串文本，请使用双直引号 ("") 或单直引号 ('') 字符。例如，以下两个字符串是等效的：

```
var str1:String = "hello";
var str2:String = 'hello';
```

您还可以使用 **new** 运算符来声明字符串，如下所示：

```
var str1:String = new String("hello");
var str2:String = new String(str1);
var str3:String = new String();           // str3 == ""
```

下面的两个字符串是等效的：

```
var str1:String = "hello";
var str2:String = new String("hello");
```

若要在使用单引号 ('') 分隔符定义的字符串文本内使用单引号 ('')，请使用反斜杠转义符 (\)。类似地，要在使用双引号 ("") 分隔符定义的字符串文本内使用双引号 ("")，请使用反斜杠转义符 (\)。下面的两个字符串是等效的：

```
var str1:String = "That's \"A-OK\"";
var str2:String = 'That\'s "A-OK"';
```

您可以根据字符串文本中存在的任何单引号或双引号来选择使用单引号或双引号，如下所示：

```
var str1:String = "ActionScript <span class='heavy'>3.0</span>";
var str2:String = '<item id="155">banana</item>';
```

请务必记住 ActionScript 可区分单直引号 ('') 和左右单引号 ('' 或 '')。对于双引号也同样如此。请使用直引号来分割字符串文本。在将文本从其它来源粘贴到 ActionScript 中时，请确保使用正确的字符。

如下表所示，可以使用反斜杠转义符 (\) 在字符串文本中定义其它字符：

转义序列	字符
\b	Backspace
\f	换页符
\n	换行符

转义序列	字符
\r	回车符
\t	Tab
\unnnn	Unicode 字符，字符代码由十六进制数字 nnnn 指定；例如，\u263a 为笑脸字符。
\xnn	ASCII 字符，字符代码由十六进制数字 nn 指定。
'	单引号
"	双引号
\	单个反斜杠字符

length 属性

每个字符串都有 length 属性，其值等于字符串中的字符数：

```
var str:String = "Adobe";
trace(str.length);           // output: 5
```

空字符串和 null 字符串的长度均为 0，如下例所示：

```
var str1:String = new String();
trace(str1.length);           // output: 0

str2:String = '';
trace(str2.length);           // output: 0
```

处理字符串中的字符

字符串中的每个字符在字符串中都有一个索引位置（整数）。第一个字符的索引位置为 0。例如，在下面的字符串中，字符 y 的位置为 0，而字符 w 的位置为 5：

```
"yellow"
```

您可以使用 charAt() 方法和 charCodeAt() 方法检查字符串各个位置上的字符，如此示例所示：

```
var str:String = "hello world!";
for (var i:int = 0; i < str.length; i++)
{
    trace(str.charAt(i), "-", str.charCodeAt(i));
}
```

在运行此代码时，会产生如下输出：

```
h - 104
e - 101
l - 108
l - 108
o - 111
- 32
w - 119
o - 111
r - 114
l - 108
d - 100
! - 33
```

您还可以通过字符代码，使用 `fromCharCode()` 方法定义字符串，如下例所示：

```
var myStr:String = String.fromCharCode(104,101,108,108,111,32,119,111,114,108,100,33);
// Sets myStr to "hello world!"
```

比较字符串

可以使用以下运算符比较字符串：<、<=、!=、==、=> 和 >。可以将这些运算符用于条件语句（如 `if` 和 `while`），如下例所示：

```
var str1:String = "Apple";
var str2:String = "apple";
if (str1 < str2)
{
    trace("A < a, B < b, C < c, ...");
}
```

在将这些运算符用于字符串时，ActionScript 会使用字符串中每个字符的字符代码值从左到右比较各个字符，如下所示：

```
trace("A" < "B"); // true
trace("A" < "a"); // true
trace("Ab" < "az"); // true
trace("abc" < "abza"); // true
```

使用 == 和 != 运算符可比较两个字符串，也可以将字符串与其它类型的对象进行比较，如下例所示：

```
var str1:String = "1";
var str1b:String = "1";
var str2:String = "2";
trace(str1 == str1b); // true
trace(str1 == str2); // false
var total:uint = 1;
trace(str1 == total); // true
```

获取其它对象的字符串表示形式

可以获取任何类型对象的字符串表示形式。所有对象都提供了 `toString()` 方法来实现此目的：

```
var n:Number = 99.47;
var str:String = n.toString();
// str == "99.47"
```

在使用 + 连接运算符连接 `String` 对象和不属于字符串的对象时，无需使用 `toString()` 方法。有关连接的详细信息，请参阅下一节。

对于给定对象，`String()` 全局函数返回的值与调用该对象的 `toString()` 方法返回的值相同。

连接字符串

字符串连接的含义是：将两个字符串按顺序合并为一个字符串。例如，可以使用 + 运算符来连接两个字符串：

```
var str1:String = "green";
var str2:String = "ish";
var str3:String = str1 + str2; // str3 == "greenish"
```

还可以使用 += 运算符来得到相同的结果，如下例所示：

```
var str:String = "green";
str += "ish"; // str == "greenish"
```

此外，String 类还包括 concat() 方法，可按如下方式对其进行使用：

```
var str1:String = "Bonjour";
var str2:String = "from";
var str3:String = "Paris";
var str4:String = str1.concat(" ", str2, " ", str3);
// str4 == "Bonjour from Paris"
```

如果使用 + 运算符（或 += 运算符）对 String 对象和非字符串的对象进行运算，则 ActionScript 会自动将非字符串对象转换为 String 对象以计算该表达式，如下例所示：

```
var str:String = "Area = ";
var area:Number = Math.PI * Math.pow(3, 2);
str = str + area; // str == "Area = 28.27433882308138"
```

但是，可以使用括号进行分组，为 + 运算符提供运算的上下文，如下例所示：

```
trace("Total: $" + 4.55 + 1.45); // output: Total: $4.551.45
trace("Total: $" + (4.55 + 1.45)); // output: Total: $6
```

在字符串中查找子字符串和模式

子字符串是字符串内的字符序列。例如，字符串 "abc" 具有如下子字符串："", "a"、"ab"、"abc"、"b"、"bc"、"c"。可使用 ActionScript 的方法来查找字符串的子字符串。

模式是在 ActionScript 中通过字符串或正则表达式定义的。例如，下面的正则表达式定义了一个特定模式，即字母 A、B 和 C 的后面跟着一个数字字符（正斜杠是正则表达式的分隔符）：

```
/ABC\d/
```

ActionScript 提供了在字符串中查找模式的方法，以及使用替换子字符串替换找到的匹配项的方法。随后的章节将介绍这些方法。

正则表达式可定义十分复杂的模式。有关详细信息，请参阅第 171 页的“[使用正则表达式](#)”。

通过字符位置查找子字符串

substr() 和 substring() 方法非常类似。两个方法都返回字符串的一个子字符串。并且两个方法都具有两个参数。在这两个方法中，第一个参数是给定字符串中起始字符的位置。不过，在 substr() 方法中，第二个参数是要返回的子字符串的长度，而在 substring() 方法中，第二个参数是子字符串的结尾处字符的位置（该字符未包含在返回的字符串中）。此示例显示了这两种方法之间的差别：

```
var str:String = "Hello from Paris, Texas!!!";
trace(str.substr(11,15)); // output: Paris, Texas!!!
trace(str.substring(11,15)); // output: Pari
```

slice() 方法与 substring() 方法的工作方式类似。当指定两个非负整数作为参数时，其运行方式将完全一样。但是，slice() 方法可以使用负整数作为参数，此时字符位置将从字符串末尾开始向前算起，如下例所示：

```
var str:String = "Hello from Paris, Texas!!!";
trace(str.slice(11,15)); // output: Pari
trace(str.slice(-3,-1)); // output: !!
trace(str.slice(-3,26)); // output: !!!
trace(str.slice(-3,str.length)); // output: !!!
trace(str.slice(-8,-3)); // output: Texas
```

可以结合使用非负整数和负整数作为 slice() 方法的参数。

查找匹配子字符串的字符位置

可以使用 `indexOf()` 和 `lastIndexOf()` 方法在字符串内查找匹配的子字符串，如下例所示：

```
var str:String = "The moon, the stars, the sea, the land";
trace(str.indexOf("the")); // output: 10
```

请注意，`indexOf()` 方法区分大小写。

可以指定第二个参数以指出在字符串中开始进行搜索的起始索引位置，如下所示：

```
var str:String = "The moon, the stars, the sea, the land"
trace(str.indexOf("the", 11)); // output: 21
```

`lastIndexOf()` 方法在字符串中查找子字符串的最后一个匹配项：

```
var str:String = "The moon, the stars, the sea, the land"
trace(str.lastIndexOf("the")); // output: 30
```

如果为 `lastIndexOf()` 方法提供了第二个参数，搜索将从字符串中的该索引位置反向（从右到左）进行：

```
var str:String = "The moon, the stars, the sea, the land"
trace(str.lastIndexOf("the", 29)); // output: 21
```

创建由分隔符分隔的子字符串数组

可使用 `split()` 方法创建子字符串数组，该数组根据分隔符进行划分。例如，可以将逗号分隔或制表符分隔的字符串分为多个字符串。

以下示例说明如何使用“与”字符（&）作为分隔符，将数组分割为多个子字符串：

```
var queryStr:String = "first=joe&last=cheng&title=manager&StartDate=3/6/65";
var params:Array = queryStr.split("&", 2); // params == ["first=joe", "last=cheng"]
```

`split()` 方法的第二个参数是可选参数，该参数定义所返回数组的最大大小。

此外，还可以使用正则表达式作为分隔符：

```
var str:String = "Give me\t5."
var a:Array = str.split(/\s+/); // a == ["Give", "me", "5."]
```

有关详细信息，请参阅第 171 页的“[使用正则表达式](#)”和《ActionScript 3.0 语言和组件参考》。

在字符串中查找模式并替换子字符串

`String` 类提供了在字符串中处理模式的以下方法：

- 使用 `match()` 和 `search()` 方法可查找与模式相匹配的子字符串。
- 使用 `replace()` 方法可查找与模式相匹配的子字符串并使用指定子字符串替换它们。

随后的章节将介绍这些方法。

您可以使用字符串或正则表达式定义在这些方法中使用的模式。有关正则表达式的详细信息，请参阅第 171 页的“[使用正则表达式](#)”。

查找匹配的子字符串

`search()` 方法返回与给定模式相匹配的第一个子字符串的索引位置，如下例所示：

```
var str:String = "The more the merrier.";
// (This search is case-sensitive.)
trace(str.search("the")); // output: 9
```

您还可以使用正则表达式定义要匹配的模式，如下例所示：

```
var pattern:RegExp = /the/i;
var str:String = "The more the merrier.";
trace(str.search(pattern)); // 0
```

`search()` 方法的输出为 0，因为字符串中第一个字符的索引位置为 0。在正则表达式中设置了 `i` 标志，因此搜索时不区分大小写。

`search()` 方法仅查找一个匹配项并返回其起始索引位置，即便在正则表达式中设置了 `g`（全局）标志。

下例展示一个更复杂的正则表达式，该表达式匹配被双引号引起的字符串：

```
var pattern:RegExp = /"[^"]*/;
var str:String = "The \"more\" the merrier.";
trace(str.search(pattern)); // output: 4

str = "The \"more the merrier.\";
trace(str.search(pattern)); // output: -1
// (Indicates no match, since there is no closing double quotation mark.)
```

`match()` 方法的工作方式与此类似。它搜索一个匹配的子字符串。但是，如果在正则表达式模式中使用了全局标志（如下例所示），`match()` 将返回一个包含匹配子字符串的数组：

```
var str:String = "bob@example.com, omar@example.org";
var pattern:RegExp = /\w*@\w*\.[org|com]+/g;
var results:Array = str.match(pattern);
```

对 `results` 数组进行如下设置：

```
["bob@example.com", "omar@example.org"]
```

有关正则表达式的详细信息，请参阅第 171 页的“[使用正则表达式](#)”第 171 页的“[使用正则表达式](#)”。

替换匹配的子字符串

您可以使用 `replace()` 方法在字符串中搜索指定模式并使用指定的替换字符串替换匹配项，如下例所示：

```
var str:String = "She sells seashells by the seashore.";
var pattern:RegExp = /sh/gi;
trace(str.replace(pattern, "sch"));
//sche sells seaschells by the seaschore.
```

请注意，在本例中，因为在正则表达式中设置了 `i` (`ignoreCase`) 标志，所以匹配的字符串是不区分大小写的；而且因为设置了 `g` (`global`) 标志，所以会替换多个匹配项。有关详细信息，请参阅第 171 页的“[使用正则表达式](#)”。

可以在替换字符串中包括以下 \$ 替换代码。下表所示的替换文本将插入并替换 \$ 替换代码：

\$ 代码	替换文本
\$\$	\$
\$&	匹配的子字符串。
\$`	字符串中位于匹配的子字符串前面的部分。此代码使用左单直引号字符 (`) 而不是单直引号 (') 或左单弯引号 (`)。
\$'	字符串中位于匹配的子字符串后的部分。此代码使用单直引号 (')。
\$n	第 n 个捕获的括号组匹配项，其中 n 是 1-9 之间的数字，而且 \$n 后面没有十进制数字。
\$nn	第 nn 个捕获的括号组匹配项，其中 nn 是一个十进制的两位数 (01-99)。如果未定义第 nn 个捕获内容，则替换文本为空字符串。

例如，下面说明了如何使用 `$2` 和 `$1` 替换代码，它们分别表示匹配的第一个和第二个捕获组：

```
var str:String = "flip-flop";
var pattern:RegExp = /(\w+)-(\w+)/g;
trace(str.replace(pattern, "$2-$1")); // flop-flip
```

也可以使用函数作为 `replace()` 方法的第二个参数。匹配的文本将被函数的返回值替换。

```

var str:String = "Now only $9.95!";
var price:RegExp = /\$(\d+\.\d+)/i;
trace(str.replace(price, usdToEuro));

function usdToEuro(matchedSubstring:String, capturedMatch1:String, index:int,
str:String):String
{
    var usd:String = capturedMatch1;
    usd = usd.replace(",","");
    var exchangeRate:Number = 0.853690;
    var euro:Number = parseFloat(usd) * exchangeRate;
    const euroSymbol:String = String.fromCharCode(8364);
    return euro.toFixed(2) + " " + euroSymbol;
}

```

在使用函数作为 `replace()` 方法的第二个形参时，将向该函数传递如下实参：

- 字符串的匹配部分。
- 任何捕获的括号组匹配项。按这种方式传递的参数数目因括号匹配项的数目而异。您可以通过检查函数代码中的 `arguments.length - 3` 来确定括号匹配项的数目。
- 字符串中匹配开始的索引位置。
- 完整的字符串。

在大小写之间转换字符串

如下例所示，`toLowerCase()` 方法和 `toUpperCase()` 方法分别将字符串中的英文字母字符转换为小写和大写：

```

var str:String = "Dr. Bob Roberts, #9."
trace(str.toLowerCase()); // dr. bob roberts, #9.
trace(str.toUpperCase()); // DR. BOB ROBERTS, #9.

```

执行完这些方法后，源字符串仍保持不变。要转换源字符串，请使用下列代码：

```
str = str.toUpperCase();
```

这些方法可处理扩展字符，而并不仅限于 `a-z` 和 `A-Z`：

```

var str:String = "José Barça";
trace(str.toUpperCase(), str.toLowerCase()); // JOSÉ BARÇA josé barça


```

示例：ASCII 字符图

此 ASCII 字符图示例说明了可以在 ActionScript 3.0 中使用 `String` 类实现的大量功能，其中包括：

- 使用 `String` 类的 `split()` 方法可从由某个字符分隔的字符串中提取值（制表符分隔的文本文件中的图像信息）。
- 使用多种字符串操作技术（包括 `split()`、连接，以及使用 `substring()` 和 `substr()` 提取字符串的一部分）可将图像标题中每个单词的第一个字母变为大写形式。
- 使用 `getCharAt()` 方法可从字符串中获取单个字符（以确定对应于某个灰度位图值的 ASCII 字符）。
- 使用字符串连接可以按一次一个字符的方式建立图像的 ASCII 字符图表示形式。

ASCII 字符图这一术语指的是图像的文本表示形式，即使用等宽字体字符（如 `Courier New` 字符）的网格来绘制图像。下图便是该应用程序所生成 ASCII 字符图的一个例子：



图形的 ASCII 字符图版本显示在右侧。

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可以在文件夹 Samples/AsciiArt 下找到 ASCIIArt 应用程序文件。该应用程序包含以下文件：

文件	描述
AsciiArtApp.mxml 或 AsciiArtApp.fla	Flash (FLA) 或 Flex (MXML) 中的主应用程序文件
com/example/programmingas3/asciiArt/AsciiArtBuilder.as	此类提供了应用程序主要功能，包括了从文本文件中提取图像元数据、加载图像和管理图像到文本的转换过程等功能。
com/example/programmingas3/asciiArt/BitmapToAsciiConverter.as	此类提供了用于将图像数据转换为字符串版本的 parseBitmapData() 方法。
com/example/programmingas3/asciiArt/Image.as	此类表示所加载的位图图像。
com/example/programmingas3/asciiArt/ImageInfo.as	此类表示 ASCII 字符图像的元数据（如标题、图像文件 URL 等）。
image/	此文件夹包含应用程序使用的图像。
txt/ImageData.txt	此制表符分隔的文本文件中包含与应用程序要加载的图像有关的信息。

提取由制表符分隔的值

此示例使用了将应用程序数据与应用程序本身分开存储的通行做法；通过这种方式，在数据发生更改时（例如，添加了另一幅图像或更改了图像标题），无需重新创建该 SWF 文件。在本例中，图像元数据（包括图像标题、实际图像文件的 URL 以及用来操作图像的某些值）存储在一个文本文件中（项目中的 txt/ImageData.txt 文件）。该文本文件的内容如下所示：

```
FILENAMETITLEWHITE_THRESHOLDBLACK_THRESHOLD
FruitBasket.jpgPear, apple, orange, and banana810
Banana.jpgA picture of a bananaC820
Orange.jpgorangeFF20
Apple.jpgpicture of an apple6E10
```

文件使用特定的制表符分隔格式。第一行为标题行。其余行包含要加载的每个位图的如下数据：

- 位图的文件名。
- 位图的显示名称。
- 位图的白色阀值和黑色阀值。这些值是十六进制值，高于这些值或低于这些值的像素将分别被认为是全白或全黑的。

应用程序启动后，**AsciiArtBuilder** 类将加载并分析文本文件的内容，以便创建它要显示的图像的“堆栈”，执行这些操作时使用 **AsciiArtBuilder** 类的 **parseImageInfo()** 方法中的如下代码：

```
var lines:Array = _imageInfoLoader.data.split("\n");
var numLines:uint = lines.length;
for (var i:uint = 1; i < numLines; i++)
{
    var imageInfoRaw:String = lines[i];
    ...
    if (imageInfoRaw.length > 0)
    {
        // Create a new image info record and add it to the array of image info.
        var imageInfo:ImageInfo = new ImageInfo();

        // Split the current line into values (separated by tab (\t)
        // characters) and extract the individual properties:
        var imageProperties:Array = imageInfoRaw.split("\t");
        imageInfo.fileName = imageProperties[0];
        imageInfo.title = normalizeTitle(imageProperties[1]);
        imageInfo.whiteThreshold = parseInt(imageProperties[2], 16);
        imageInfo.blackThreshold = parseInt(imageProperties[3], 16);
        result.push(imageInfo);
    }
}
```

文本文件的完整内容包含在单个 **String** 实例中，即 **_imageInfoLoader.data** 属性。使用 **split()** 方法并以换行符 (**\n**) 为参数，将该 **String** 实例分割到一个 **Array** (**lines**) 中，数组元素为文本文件的各个行。然后，代码使用一个循环处理每行（第一行除外，因为它只包含标题而不包含实际内容）。在循环内部，再一次使用 **split()** 方法将每行的内容分割到一组值（名为 **imageProperties** 的 **Array** 对象）中。在本例中，用于 **split()** 方法的参数为制表符 (**\t**)，因为每行中的值均由制表符进行分隔。

使用 **String** 的方法标准化图像标题

此应用程序的设计目标之一便是使用一种标准格式显示所有图像标题，即标题中每一个单词的第一个字母均为大写形式（在英文标题中通常不大写的少数单词除外）。应用程序并不假定文本文件已经包含格式正确的标题，它在从文本文件中提取标题时对标题格式进行设置。

在前面的代码清单中，使用了如下代码行来提取每个图像元数据值：

```
imageInfo.title = normalizeTitle(imageProperties[1]);
```

在该代码中，来自文本文件的图像标题在存储到 **ImageInfo** 对象之前先传递给 **normalizeTitle()** 方法进行处理：

```
private function normalizeTitle(title:String):String
{
    var words:Array = title.split(" ");
    var len:uint = words.length;
    for (var i:uint; i < len; i++)
    {
        words[i] = capitalizeFirstLetter(words[i]);
    }

    return words.join(" ");
}
```

此方法使用 **split()** 方法将标题分割为各个单独的单词（由空格字符加以分隔），然后将每个单词传递给 **capitalizeFirstLetter()** 方法进行处理，接着使用 **Array** 类的 **join()** 方法将单词重新合并为一个字符串。

如同其名称所表达的含义，**capitalizeFirstLetter()** 方法实际执行将每个单词的第一个字母变为大写形式的工作：

```
/*
 * Capitalizes the first letter of a single word, unless it's one of
 * a set of words that are normally not capitalized in English.
 */
private function capitalizeFirstLetter(word:String):String
{
    switch (word)
    {
        case "and":
        case "the":
        case "in":
        case "an":
        case "or":
        case "at":
        case "of":
        case "a":
            // Don't do anything to these words.
            break;
        default:
            // For any other word, capitalize the first character.
            var firstLetter:String = word.substr(0, 1);
            firstLetter = firstLetter.toUpperCase();
            var otherLetters:String = word.substring(1);
            word = firstLetter + otherLetters;
    }
    return word;
}
```

在英文中，如果标题中某个单词为以下单词之一，则不会将其首字符变为大写形

式：“and”、“the”、“in”、“an”、“or”、“at”、“of”或“a”。（这是相关规则的简化版本。）为了实现此逻辑，代码首先使用 switch 语句来检查单词是否为不应将其首字符大写的单词之一。如果是，代码直接跳出 switch 语句。另一方面，如果单词应大写，则在几个步骤中完成此操作，如下所示：

- 1 使用 substr(0, 1) 提取出单词的第一个字母，该命令从位于索引位置 0 的字符（即字符串的第一个字母，由第一个参数 0 指定）开始提取子字符串。该子字符串的长度为一个字符（由第二个参数 1 指定）。
- 2 使用 toUpperCase() 方法将该字符变为大写形式。
- 3 使用 substring(1) 提取原始单词的其余字符，该命令提取从索引位置 1（第二个字母）开始直至字符串结尾（通过将 substring() 方法的第二个参数保留为空进行指定）的子字符串。
- 4 使用字符串连接 firstLetter + otherLetters 将刚才变为大写形式的第一个字母与其余字母合并在一起，创建出最终的单词。

生成 ASCII 字符图文本

BitmapToAsciiConverter 类提供了将位图图像转换为其 ASCII 文本表示形式的功能。此过程由 parseBitmapData() 方法执行，下面展示了该方法的部分工作过程：

```
var result:String = "";

// Loop through the rows of pixels top to bottom:
for (var y:uint = 0; y < _data.height; y += verticalResolution)
{
    // Within each row, loop through pixels left to right:
    for (var x:uint = 0; x < _data.width; x += horizontalResolution)
    {
        ...

        // Convert the gray value in the 0-255 range to a value
        // in the 0-64 range (since that's the number of "shades of
        // gray" in the set of available characters):
        index = Math.floor(grayVal / 4);
        result += palette.charAt(index);

    }
    result += "\n";
}
return result;
```

此代码首先定义一个名为 `result` 的 `String` 实例，用于构建位图图像的 ASCII 字符图版本。然后，它遍历源位图图像的每个像素。通过使用若干颜色处理技术（为了简便起见，此处省略了对这些技术的介绍），它将每个像素的红色、绿色和蓝色值转换为单个灰度值（一个介于 0-255 之间的数字）。接着，代码将该值除以 4（如下所示）以将其转换为介于 0-63 之间的一个值，此值存储在变量 `index` 中。（之所以使用 0-63 的范围，是因为此应用程序使用的可用 ASCII 字符的“调色板”包含 64 个值。）该字符调色板在 `BitmapToAsciiConverter` 类中定义为一个 `String` 实例：

```
// The characters are in order from darkest to lightest, so that their
// position (index) in the string corresponds to a relative color value
// (0 = black).
private static const palette:String = "@#$%&8BMW*mwqpdःkhaQ0OZXYUJCLtfjzxnuvcr[]{}1()|/?Il!i><+_~-;,. ";
```

因为变量 `index` 定义调色板中的哪个 ASCII 字符对应于位图图像中的当前像素，所以可使用 `charAt()` 方法从 `palette` 字符串中检索该字符。然后，使用连接赋值运算符 (`+=`) 将其追加到 `result` 字符串实例。此外，在每行像素的末尾，会将一个换行符连接到 `result` 字符串的末尾，强制该行换行以创建新的一行字符“像素”。

第 8 章：处理数组

使用数组可以在单数据结构中存储多个值。可以使用简单的索引数组（使用固定有序整数索引存储值），也可以使用复杂的关联数组（使用任意键存储值）。数组也可以是多维的，即包含本身是数组的元素。最后，您可以对其元素均是同一数据类型的实例的数组使用 **Vector**。本章讨论如何创建和操作各种类型的数组。

数组基础知识

处理数组简介

在编程中经常需要处理一组项目而不是单个对象。例如在一个音乐播放器应用程序中，您可能需要创建一个待播放歌曲列表。您不希望必须为该列表中的每首歌曲都单独创建一个变量。更好的做法是将所有 **Song** 对象放在一个包中，从而能够将其作为一个组进行处理。

数组是一种编程元素，它用作一组项目的容器，如一组歌曲。通常，数组中的所有项目都是相同类的实例，但这在 **ActionScript** 中并不是必需的。数组中的各个项目称为数组的“元素”。可以将数组视为变量的“文件柜”。变量可以作为元素添加到数组中，就像将文件夹放到文件柜中一样。您可以将数组作为单个变量使用（就像将整个文件柜搬到另一个地方一样）。您可以将变量作为一个组使用（就像逐一浏览文件夹以搜索一条信息一样）。也可以分别访问这些变量（就像打开文件柜并选择一个文件夹一样）。

例如，假设您要创建一个音乐播放器应用程序，用户可以在其中选择多首歌曲，并将这些歌曲添加到播放列表中。在您的 **ActionScript** 代码中有一个名为 **addSongsToPlaylist()** 的方法，该方法接受单个数组作为参数。无论要将多少首歌曲（几首、很多首甚至只有一首）添加到列表中，您都只调用一次 **addSongsToPlaylist()** 方法，并向其传递包含 **Song** 对象的数组。在 **addSongsToPlaylist()** 方法中，可以使用循环来逐个访问数组元素（歌曲），并将歌曲实际添加到播放列表中。

最常见的 **ActionScript** 数组类型为“索引数组”。在索引数组中，每个项目都存储在编号位置（称为“索引”）。可以使用该编号来访问项目，如同地址一样。索引数组可以很好地满足大多数编程需要。**Array** 类是用于表示索引数组的常见类。

索引数组常用于存储具有相同类型的多个项目（作为同一类的实例的对象）。**Array** 类没有任何办法限制它所包含的项目的类型。**Vector** 类是一种索引数组类型，其中单个数组中的所有项目都具有同一类型。使用 **Vector** 实例而不是 **Array** 实例还可以提供性能改进和其它优势。从 **Flash Player 10** 和 **Adobe AIR 1.5** 开始提供 **Vector** 类。

索引数组的一个特殊用法是“多维数组”。多维数组是一种索引数组，其中的元素也是索引数组（这些数组又包含其它元素）。

另一种数组类型是“关联数组”，该数组使用字符串“键”（而不是数字索引）来标识各个元素。最后，**ActionScript 3.0** 还包括表示“字典”的 **Dictionary** 类。字典是允许您将任何类型的对象用作键来区分元素的数组。

常见数组任务

本章介绍了以下常见的数组使用活动：

- 使用 **Array** 类和 **Vector** 类创建索引数组
- 添加和删除数组元素
- 对数组元素进行排序
- 提取数组的某些部分
- 处理关联数组和字典
- 处理多维数组
- 复制数组元素

- 创建数组子类

重要概念和术语

以下参考列表包含将会在本章中遇到的重要术语：

- 数组：用作容器以将多个对象组合在一起的对象
- 数组访问 ([]) 运算符：一对方括号，其中含有唯一标识数组元素的索引或键。此语法用在数组变量名称之后，以指定数组的单个元素而不是整个数组。
- 关联数组：使用字符串键来标识各个元素的数组
- 基本类型：允许 **Vector** 实例存储的对象的数据类型
- 字典：其项目由一对对象（称为键和值）组成的数组。它使用键来标识单个元素，而不是使用数字索引。
- 元素：数组中的单个项目
- 索引：用于标识索引数组中的单个元素的数字“地址”
- 索引数组：标准类型的数组，将每个元素存储在编号位置中，并使用数字（索引）来标识各个元素
- 键：用于标识关联数组或字典中的单个元素的字符串或对象
- 多维数组：包含的项目是数组（而不是单个值）的数组
- T：本文档中使用的标准约定，用于表示 **Vector** 实例的基本类型（与具体的基本类型无关）。T 约定用于表示类名称，如 Type 参数说明中所示。（“T”表示“类型”，如同在“数据类型”中一样。）
- Type 参数：与 **Vector** 类名称一起使用以指定 **Vector** 的基本类型（它存储的对象的数据类型）的语法。该语法包括一个点(.)，然后是由尖括号(<>)括起来的数据类型名称。放在一起后类似于：**Vector.<T>**。在本文档中，在 type 参数中指定的类通常表示为 T。
- 矢量：一种数组类型，其所有元素都是同一数据类型的实例

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。实质上本章中的代码清单包括适当的 **trace()** 函数调用。要测试本章中的代码清单，请执行以下操作：

- 1 使用 Flash 创作工具创建空文档
- 2 在时间轴上选择一个关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 通过“控制”>“测试影片”运行程序。

可在“输出”面板中看到 **trace()** 函数的结果。

用于测试示例代码清单的此项技术以及其它技术在第 29 页的“[测试本章内的示例代码清单](#)”中有详细的介绍。

索引数组

索引数组存储一系列经过组织的单个或多个值，其中的每个值都可以通过使用一个无符号整数值进行访问。第一个索引始终是数字 0，且添加到数组中的每个后续元素的索引以 1 为增量递增。在 ActionScript 3.0 中，有两个类可用作索引数组：**Array** 类和 **Vector** 类。

索引数组使用无符号 32 位整数作为索引号。索引数组的最大大小为 $2^{32} - 1$ ，即 4,294,967,295。如果尝试创建超过该最大大小的数组，则会导致运行时错误。

若要访问索引数组中的单个元素，请使用数组访问 ([]) 运算符指定要访问的元素的索引位置。例如，下面的代码表示名为 songTitles 的索引数组中的第一个元素（位于索引 0 处的元素）：

```
songTitles[0]
```

数组变量名称后跟索引（在方括号中）的组合用作一个标识符。（换句话说，可以按照变量名称的任何使用方式来使用这个组合）。可以在赋值语句左侧使用名称和索引来向索引数组元素赋值。

```
songTitles[1] = "Symphony No. 5 in D minor";
```

同样，可以在赋值语句右侧使用名称和索引来检索索引数组元素的值。

```
var nextSong:String = songTitles[2];
```

您还可以在方括号中使用变量而不是提供显式值。（变量必须包含一个非负整数值，如 uint、正 int 或正整数 Number 实例）。此方法通常用于“循环访问”索引数组中的元素，并对某些元素或所有元素执行操作。下面的代码清单演示了这种方法。这段代码使用循环来访问名为 oddNumbers 的 Array 对象中的每个值。它使用 trace() 语句以“oddNumber[index] = value”的形式输出每个值：

```
var oddNumbers:Array = [1, 3, 5, 7, 9, 11];
var len:uint = oddNumbers.length;
for (var i:uint = 0; i < len; i++)
{
    trace("oddNumbers[" + i.toString() + "] = " + oddNumbers[i].toString());
}
```

Array 类

索引数组的第一个类型为 **Array** 类。**Array** 实例的值可以为任意数据类型。同一 **Array** 对象中包含的对象可以具有不同的数据类型。例如，一个 **Array** 实例可以在索引 0 处有 String 值，在索引 1 处有 Number 实例，而在索引 2 处有 XML 对象。

Vector 类

ActionScript 3.0 中可用的另一种索引数组类型为 **Vector** 类。**Vector** 实例是“指定类型的数组”，这表示 **Vector** 实例中的所有元素始终具有同一数据类型。

注：从 Flash Player 10 和 Adobe AIR 1.5 开始提供 **Vector** 类。

在声明 **Vector** 变量或实例化 **Vector** 对象时，要显式指定 **Vector** 可以包含的对象的数据类型。指定的数据类型称为 **Vector** 的“基本类型”。在运行时和编译时（在严格模式下），会检查任何设置 **Vector** 元素的值或从 **Vector** 检索值的代码。如果要添加或检索的对象的数据类型与 **Vector** 的基本类型不匹配，则会发生错误。

除数据类型限制之外，**Vector** 类还具有其他一些限制，从而有别于 **Array** 类：

- **Vector** 是一种密集数组。即使某个 **Array** 对象在位置 1 到 6 没有值，该对象的索引 0 和 7 处也可以有值。但是，**Vector** 的每个索引位置都必须有值（或为 null）。
- **Vector** 还可以是固定长度。这表示 **Vector** 包含的元素数不能更改。
- 对 **Vector** 的元素的访问需要接受范围检查。绝对不能从大于最后一个元素索引 (length - 1) 的索引中读取值。绝对不能对超过当前最后一个索引一个以上位置的索引设置值（也就是说，只能在现有索引或索引 [length] 处设置值）。

由于 **Vector** 具有这些限制，因此 **Vector** 相对于所有元素均为单个类的实例的 **Array** 实例有两个主要优点：

- **性能：**使用 **Vector** 实例时的数组元素访问和迭代的速度比使用 **Array** 实例时的速度要快很多。
- **类型安全性：**在严格模式下，编译器可以识别数据类型错误。这类错误的例子包括将数据类型错误的值分配给 **Vector** 或从 **Vector** 中读取值时使用错误的数据类型。在运行时，当向 **Vector** 对象添加数据或从 **Vector** 对象读取数据时也会检查数据类型。但请注意，当使用 **push()** 方法或 **unshift()** 方法向 **Vector** 添加值时，在编译时不会检查参数的数据类型。不过在使用这些方法时，仍会在运行时检查值。

除了有一些限制和优点以外，**Vector** 类与 **Array** 类非常相似。**Vector** 对象的属性和方法与 **Array** 的属性和方法类似（大多数情况下完全相同）。对于任何需要使用所有元素都具有相同数据类型的 **Array** 的情况，**Vector** 实例都更为可取。

创建数组

可以使用多种方法来创建 **Array** 实例或 **Vector** 实例。但是，创建每种数组类型的方法多少有些不同。

创建 **Array** 实例

可以通过调用 **Array()** 构造函数或使用 **Array** 文本语法来创建 **Array** 对象。

Array() 构造函数有三种使用方式。第一种，如果调用不带参数的构造函数，会得到空数组。可以使用 **Array** 类的 **length** 属性来验证数组是否不包含元素。例如，下面的代码调用不带参数的 **Array()** 构造函数：

```
var names:Array = new Array();
trace(names.length); // output: 0
```

第二种，如果将一个数字用作 **Array()** 构造函数的唯一参数，则会创建长度等于此数值的数组，并且每个元素的值都设置为 **undefined**。参数必须为介于值 0 和 4,294,967,295 之间的无符号整数。例如，下面的代码调用带有一个数字参数的 **Array()** 构造函数：

```
var names:Array = new Array(3);
trace(names.length); // output: 3
trace(names[0]); // output: undefined
trace(names[1]); // output: undefined
trace(names[2]); // output: undefined
```

第三种，如果调用构造函数并传递一个元素列表作为参数，将创建具有与每个参数对应的元素的数组。下面的代码将三个参数传递给 **Array()** 构造函数：

```
var names:Array = new Array("John", "Jane", "David");
trace(names.length); // output: 3
trace(names[0]); // output: John
trace(names[1]); // output: Jane
trace(names[2]); // output: David
```

也可以使用 **Array** 文本创建数组。可以将 **Array** 文本直接分配给数组变量，如下面的示例所示：

```
var names:Array = ["John", "Jane", "David"];
```

创建 **Vector** 实例

可以通过调用 **Vector.<T>()** 构造函数创建 **Vector** 实例。还可以通过调用 **Vector.<T>()** 全局函数创建 **Vector**。该函数将指定对象转换为 **Vector** 实例。ActionScript 没有等效于 **Array** 文本语法的 **Vector**。

只要声明 **Vector** 变量（**Vector** 方法参数或方法返回类型也一样），就需要指定 **Vector** 变量的基本类型。在通过调用 **Vector.<T>()** 构造函数创建 **Vector** 实例时也要指定基本类型。换句话说，只要在 ActionScript 中使用术语 **Vector**，就需要指定基本类型。

可以使用 **type** 参数语法指定 **Vector** 的基本类型。在代码中，**type** 参数紧跟单词 **Vector**。它包括一个点（**.**），然后是由尖括号（**<>**）括起来的基类名称，如此示例中所示：

```
var v:Vector.<String>;
v = new Vector.<String>();
```

在此示例的第一行内容中，变量 **v** 声明为 **Vector.<String>** 实例。换句话说，它表示只能包含 **String** 实例的索引数组。第二行调用 **Vector()** 构造函数创建一个具有相同 **Vector** 类型的实例（也就是其中所有元素都是 **String** 对象的 **Vector**）。该示例将该对象分配给 **v**。

使用 **Vector.<T>()** 构造函数

如果使用不带任何参数的 **Vector.<T>()** 构造函数，则该函数会创建一个空的 **Vector** 实例。可以通过检查 **Vector** 的 **length** 属性来测试它是否为空。例如，下面的代码调用不带任何参数的 **Vector.<T>()** 构造函数：

```
var names:Vector.<String> = new Vector.<String>();
trace(names.length); // output: 0
```

如果您预先知道 **Vector** 最初需要多少元素，则可以预定义 **Vector** 中的元素数。若要使用特定数量的元素创建 **Vector**，请将元素数作为第一个参数（**length** 参数）进行传递。因为 **Vector** 元素不能为空，所以会使用具有基本类型的实例填充这些元素。如果基本类型是允许使用 **null** 值的引用类型，则所有元素都包含 **null**。否则，所有元素都包含该类的默认值。例如，**uint** 变量不能为 **null**。因此，在下面的代码清单中，使用七个元素创建名为 **ages** 的 **Vector**，其中每个元素都包含值 0：

```
var ages:Vector.<uint> = new Vector.<uint>(7);
trace(ages); // output: 0,0,0,0,0,0,0,
```

最后，还可以使用 **Vector.<T>()** 构造函数创建固定长度 **Vector**，方法是将 **true** 作为第二个参数（**fixed** 参数）进行传递。在这种情况下，将使用指定的元素数创建 **Vector**，且元素数不可更改。但是请注意，仍然可以更改固定长度 **Vector** 的元素值。

与 **Array** 类不同，您不能将值列表传递给 **Vector.<T>()** 构造函数来指定 **Vector** 的初始值。

使用 **Vector.<T>()** 全局函数

除了 **Vector.<T>()** 构造函数之外，还可以使用 **Vector.<T>()** 全局函数创建 **Vector** 对象。**Vector.<T>()** 全局函数是一个转换函数。当调用 **Vector.<T>()** 全局函数时，需指定该方法返回的 **Vector** 的基本类型。可将单个索引数组（**Array** 或 **Vector** 实例）作为参数进行传递。该方法随后返回具有指定基本类型的 **Vector**，其中包含源数组参数中的值。下面的代码清单演示了用于调用 **Vector.<T>()** 全局函数的语法：

```
var friends:Vector.<String> = Vector.<String>(["Bob", "Larry", "Sarah"]);
```

Vector.<T>() 全局函数在两个级别上执行数据类型转换。首先，当 **Array** 实例传递给该函数时，会返回 **Vector** 实例。其次，无论源数组是 **Array** 还是 **Vector** 实例，该函数都会试图将源数组的元素转换为基本类型的值。该转换使用标准 ActionScript 数据类型转换规则。例如，下面的代码清单将源 **Array** 中的 **String** 值转换为结果 **Vector** 中的整数。第一个值（“1.5”）的小数部分被截断，非数字的第三个值（“Waffles”）在结果中转换为 0：

```
var numbers:Vector.<int> = Vector.<int>("1.5", "17", "Waffles");
trace(numbers); // output: 1,17,0
```

如果无法转换任何源元素，则会发生错误。

当代码调用 **Vector.<T>()** 全局函数时，如果源数组中的某个元素是指定基本类型的子类的实例，则该元素会添加到结果 **Vector** 中（不发生错误）。只有使用 **Vector.<T>()** 全局函数才能将基本类型为 **T** 的 **Vector** 转换为基本类型为 **T** 的超类的 **Vector**。

插入数组元素

用于将元素添加到索引数组的最基本方法是使用数组访问（`[]`）运算符。若要设置索引数组元素的值，请在赋值语句的左侧使用 **Array** 或 **Vector** 对象名称和索引编号：

```
songTitles[5] = "Happy Birthday";
```

如果 **Array** 或 **Vector** 在该索引处还没有元素，则会创建该索引并将值存储在那里。如果该索引处存在值，则新值会替换现有值。

Array 对象允许您在任何索引位置创建元素。但是对于 **Vector** 对象，您只能向现有索引或下一个可用索引赋值。下一个可用索引对应于 **Vector** 对象的 **length** 属性。向 **Vector** 对象添加新元素的最安全方式是使用类似于下面的清单的代码：

```
myVector[myVector.length] = valueToAdd;
```

可以使用 **Array** 和 **Vector** 类的三种方法（**push()**、**unshift()** 和 **splice()**）将元素插入索引数组。**push()** 方法用于在数组末尾添加一个或多个元素。换言之，使用 **push()** 方法在数组中插入的最后一个元素将具有最大索引号。**unshift()** 方法用于在数组开头插入一个或多个元素，并且始终在索引号 0 处插入。**splice()** 方法用于在数组中的指定索引处插入任意数目的项目。

下面的示例对所有三种方法进行了说明。它创建一个名为 **planets** 的数组，以便按照行星距离太阳由近到远的顺序存储各个行星的名称。首先，调用 **push()** 方法以添加初始项目 **Mars**。接着，调用 **unshift()** 方法在数组开头插入项 **Mercury**。最后，调用 **splice()** 方法在 **Mercury** 之后和 **Mars** 之前插入项 **Venus** 和 **Earth**。传递给 **splice()** 的第一个参数是整数 1，它用于指示从索引 1 处开始插入。传递给 **splice()** 的第二个参数是整数 0，它表示不应删除任何项。传递给 **splice()** 的第三和第四个参数 **Venus** 和 **Earth** 为要插入的项。

```
var planets:Array = new Array();
planets.push("Mars"); // array contents: Mars
planets.unshift("Mercury"); // array contents: Mercury,Mars
planets.splice(1, 0, "Venus", "Earth");
trace(planets); // array contents: Mercury,Venus,Earth,Mars
```

`push()` 和 `unshift()` 方法均返回一个无符号整数，它们表示修改后的数组长度。在用于插入元素时，`splice()` 方法返回空数组，这看上去也许有点奇怪，但考虑到 `splice()` 方法的多用途性，您便会觉得这样更有意义。通过使用 `splice()` 方法，不仅可以将元素插入到数组中，而且还可以从数组中删除元素。用于删除元素时，`splice()` 方法将返回包含被删除元素的数组。

注：如果某个 `Vector` 对象的 `fixed` 属性为 `true`，则不能更改该 `Vector` 中的元素数。如果尝试使用此处介绍的方法向固定长度 `Vector` 添加新元素，则会发生错误。

检索值和删除数组元素

用于检索索引数组中的元素值的最简单方法是使用数组访问 (`[]`) 运算符。若要检索索引数组元素的值，请在赋值语句的右侧使用 `Array` 或 `Vector` 对象名称和索引编号：

```
var myFavoriteSong:String = songTitles[3];
```

可以尝试使用不存在元素的位置的索引来检索 `Array` 或 `Vector` 中的值。在这种情况下，`Array` 对象返回未定义的值，而 `Vector` 会引发 `RangeError` 异常。

可以使用 `Array` 和 `Vector` 类的三种方法 (`pop()`、`shift()` 和 `splice()`) 删除元素。`pop()` 方法用于从数组末尾删除一个元素。换言之，它将删除位于最大索引号处的元素。`shift()` 方法用于从数组开头删除一个元素，也就是说，它始终删除索引号 0 处的元素。`splice()` 方法既可用来插入元素，也可以删除任意数目的元素，其操作的起始位置位于由发送到此方法的第一个参数指定的索引号处。

下面的示例使用所有三种方法从 `Array` 实例中删除元素。该示例创建一个名为 `oceans` 的 `Array`，用于存储较大水域的名称。`Array` 中的某些名称为湖泊的名称而非海洋的名称，因此需要将其删除。

首先，使用 `splice()` 方法删除项 `Aral` 和 `Superior`，并插入项 `Atlantic` 和 `Indian`。传递给 `splice()` 的第一个参数是整数 2，它表示应从列表中的第三个项（即索引 2 处）开始执行操作。第二个参数 2 表示应删除两个项。其余两个参数 `Atlantic` 和 `Indian` 是要在索引 2 处插入的值。

然后，使用 `pop()` 方法删除数组中的最后一个元素 `Huron`。最后，使用 `shift()` 方法删除数组中的第一个项 `Victoria`。

```
var oceans:Array = ["Victoria", "Pacific", "Aral", "Superior", "Indian", "Huron"];
oceans.splice(2, 2, "Arctic", "Atlantic"); // replaces Aral and Superior
oceans.pop(); // removes Huron
oceans.shift(); // removes Victoria
trace(oceans); // output: Pacific,Arctic,Atlantic,Indian
```

`pop()` 和 `shift()` 方法均返回已删除的项。对于 `Array` 实例，由于数组可以包含任意数据类型的值，因而返回值的数据类型为 `Object`。对于 `Vector` 实例，返回值的数据类型是 `Vector` 的基本类型。`splice()` 方法返回包含所删除的值的 `Array` 或 `Vector`。可以更改 `oceans` `Array` 示例，以使 `splice()` 调用将返回的 `Array` 分配给新的 `Array` 变量，如下面的示例所示：

```
var lakes:Array = oceans.splice(2, 2, "Arctic", "Atlantic");
trace(lakes); // output: Aral,Superior
```

您可能会遇到对 `Array` 对象元素使用 `delete` 运算符的代码。`delete` 运算符会将 `Array` 元素的值设置为 `undefined`，但它不会从 `Array` 中删除元素。例如，下面的代码对 `oceans` `Array` 中的第三个元素使用 `delete` 运算符，但此 `Array` 的长度仍然为 5：

```
var oceans:Array = ["Arctic", "Pacific", "Victoria", "Indian", "Atlantic"];
delete oceans[2];
trace(oceans); // output: Arctic,Pacific,,Indian,Atlantic
trace(oceans[2]); // output: undefined
trace(oceans.length); // output: 5
```

可以使用数组的 `length` 属性截断 `Array` 或 `Vector`。如果您将某个索引数组的 `length` 属性设置为小于该数组的当前长度的长度，则该数组会被截断，从而删除存储在比 `length` 的新值减 1 大的索引编号处的所有元素。例如，如果 `oceans` 数组进行了排序，以使所有有效项都位于数组开头，则可以使用 `length` 属性删除位于数组后部的项，如下面的代码所示：

```
var oceans:Array = ["Arctic", "Pacific", "Victoria", "Aral", "Superior"];
oceans.length = 2;
trace(oceans); // output: Arctic,Pacific
```

注 如果某个 `Vector` 对象的 `fixed` 属性为 `true`，则不能更改该 `Vector` 中的元素数。如果尝试使用此处介绍的方法删除固定长度 `Vector` 中的元素或截断固定长度 `Vector`，则会发生错误。

对数组排序

可以使用三种方法（`reverse()`、`sort()` 和 `sortOn()`）通过排序或反向排序来更改索引数组的顺序。所有这些方法都用来修改现有数组。下表概述了这些方法及其针对 `Array` 和 `Vector` 对象的行为：

方法	<code>Array</code> 行为	<code>Vector</code> 行为
<code>reverse()</code>	更改元素的顺序，使最后一个元素变为第一个元素，倒数第二个元素变为第二个元素，依此类推。	与 <code>Array</code> 行为相同
<code>sort()</code>	用于按照各种预定义的方式对 <code>Array</code> 的元素进行排序（如字母顺序或数字顺序）。还可以指定自定义排序算法。	根据您指定的自定义排序算法对元素进行排序
<code>sortOn()</code>	用于对具有一个或多个公共属性的对象进行排序，排序时指定这样的属性作为排序键	在 <code>Vector</code> 类中不可用

`reverse()` 方法

`reverse()` 方法不带参数，也不返回值，但可以将数组从当前顺序切换为相反顺序。以下示例颠倒了 `oceans` 数组中列出的海洋顺序：

```
var oceans:Array = ["Arctic", "Atlantic", "Indian", "Pacific"];
oceans.reverse();
trace(oceans); // output: Pacific,Indian,Atlantic,Arctic
```

使用 `sort()` 方法的基本排序（仅适用于 `Array` 类）

对于 `Array` 实例，`sort()` 方法按照“默认排序顺序”重新排列数组中的元素。默认排序顺序具有以下特征：

- 排序区分大小写，也就是说大写字符优先于小写字符。例如，字母 D 优先于字母 b。
- 排序按照升序进行，也就是说低位字符代码（例如 A）优先于高位字符代码（例如 B）。
- 排序将相同的值互邻放置，并且不区分顺序。
- 排序基于字符串，也就是说，在比较元素之前，先将其转换为字符串（例如，10 优先于 3，因为相对于字符串 "3" 而言，字符串 "1" 具有低位字符代码）。

您也许需要不区分大小写或者按照降序对 `Array` 进行排序，或者您的数组中包含数字，从而需要按照数字顺序而非字母顺序进行排序。`Array` 类的 `sort()` 方法具有 `options` 参数，可通过该参数改变默认排序顺序的各个特征。`options` 是由 `Array` 类中的一组静态常量定义的，如以下列表所示：

- `Array.CASEINSENSITIVE`: 此选项可使排序不区分大小写。例如，小写字母 b 优先于大写字母 D。
- `Array.DESCENDING`: 用于颠倒默认的升序排序。例如，字母 B 优先于字母 A。
- `Array.UNIQUESORT`: 如果发现两个相同的值，此选项将导致排序中止。
- `Array.NUMERIC`: 这会导致排序按照数字顺序进行，比方说 3 优先于 10。

以下示例重点说明了这些选项中的某些选项。它创建一个名为 `poets` 的 `Array`，并使用几种不同的选项对其进行排序。

```

var poets:Array = ["Blake", "cummings", "Angelou", "Dante"];
poets.sort(); // default sort
trace(poets); // output: Angelou,Blake,Dante,cummings

poets.sort(Array.CASEINSENSITIVE);
trace(poets); // output: Angelou,Blake,cummings,Dante

poets.sort(Array.DESCENDING);
trace(poets); // output: Cummings,Dante,Blake,Angelou

poets.sort(Array.DESCENDING | Array.CASEINSENSITIVE); // use two options
trace(poets); // output: Dante,Cummings,Blake,Angelou

```

使用 **sort()** 方法的自定义排序（适用于 **Array** 和 **Vector** 类）

除了可用于 **Array** 对象的基本排序之外，您还可以定义自定义排序规则。此方法是可用于 **Vector** 类的唯一一种形式的 **sort()** 方法。若要定义自定义排序，请编写自定义排序函数，并将该函数作为参数传递给 **sort()** 方法。

例如，如果有一个名称列表，其中每个列表元素都包含一个人的全名，但现在要按照姓氏对列表排序，则必须使用自定义排序函数分析每个元素，并在排序函数中使用姓氏。下面的代码说明如何使用作为参数传递给 **Array.sort()** 方法的自定义函数来完成上述工作：

```

var names:Array = new Array("John Q. Smith", "Jane Doe", "Mike Jones");
function orderLastName(a, b):int
{
    var lastName:RegExp = /\b\S+\$/;
    var name1 = a.match(lastName);
    var name2 = b.match(lastName);
    if (name1 < name2)
    {
        return -1;
    }
    else if (name1 > name2)
    {
        return 1;
    }
    else
    {
        return 0;
    }
}
trace(names); // output: John Q. Smith, Jane Doe, Mike Jones
names.sort(orderLastName);
trace(names); // output: Jane Doe, Mike Jones, John Q. Smith

```

自定义排序函数 **orderLastName()** 使用正则表达式从每个元素中提取姓，以用于比较操作。针对 **names** 数组调用 **sort()** 方法时，函数标识符 **orderLastName** 用作唯一的参数。排序函数接受两个参数 **a** 和 **b**，因为它每次对两个数组元素进行操作。排序函数的返回值指示应如何对元素排序：

- 返回值 **-1** 表示第一个参数 **a** 优先于第二个参数 **b**。
- 返回值 **1** 表示第二个参数 **b** 优先于第一个参数 **a**。
- 返回值为 **0** 表示元素具有相同的排序优先级。

sortOn() 方法（仅适用于 **Array** 类）

sortOn() 方法是为具有包含对象的元素的 **Array** 对象设计的。这些对象应至少具有一个可用作排序键的公共属性。如果将 **sortOn()** 方法用于任何其它类型的数组，则会产生意外结果。

注：**Vector** 类不包含 **sortOn()** 方法。此方法仅用于 **Array** 对象。

下面的示例修改 **poets** **Array**，以使每个元素均为对象而非字符串。每个对象既包含诗人的姓氏又包含诗人的出生年份。

```
var poets:Array = new Array();
poets.push({name:"Angelou", born:"1928"});
poets.push({name:"Blake", born:"1757"});
poets.push({name:"cummings", born:"1894"});
poets.push({name:"Dante", born:"1265"});
poets.push({name:"Wang", born:"701"});
```

可以使用 `sortOn()` 方法，按照 `born` 属性对 `Array` 进行排序。`sortOn()` 方法定义两个参数 `fieldName` 和 `options`。必须将 `fieldName` 参数指定为字符串。在以下示例中，使用两个参数 `"born"` 和 `Array.NUMERIC` 来调用 `sortOn()`。`Array.NUMERIC` 参数用于确保按照数字顺序进行排序，而不是按照字母顺序。即使所有数字具有相同的数位，这也是一种很好的做法，因为当后来在数组中添加较少数位或较多数位的数字时，它会确保排序如期继续进行。

```
poets.sortOn("born", Array.NUMERIC);
for (var i:int = 0; i < poets.length; ++i)
{
    trace(poets[i].name, poets[i].born);
}
/* output:
Wang 701
Dante 1265
Blake 1757
cummings 1894
Angelou 1928
*/
```

在不修改原始数组的情况下进行排序（仅适用于 `Array` 类）

通常，`sort()` 和 `sortOn()` 方法用于修改 `Array`。如果要对 `Array` 排序而又不修改现有数组，请将 `Array.RETURNINDEXEDARRAY` 常量作为 `options` 参数的一部分进行传递。此选项将指示方法返回反映排序的新 `Array`，同时保留原始 `Array` 原封不动。方法返回的 `Array` 为由反映新排序顺序的索引号组成的简单 `Array`，不包含原始 `Array` 的任何元素。例如，若要根据出生年份对 `poets` `Array` 排序而不修改该 `Array`，请在为 `options` 形参传递的实参中包括 `Array.RETURNINDEXEDARRAY` 常量。

下面的示例将返回的索引信息存储在名为 `indices` 的 `Array` 中，然后使用 `indices` 数组和未修改的 `poets` 数组按出生年份的顺序输出诗人：

```
var indices:Array;
indices = poets.sortOn("born", Array.NUMERIC | Array.RETURNINDEXEDARRAY);
for (var i:int = 0; i < indices.length; ++i)
{
    var index:int = indices[i];
    trace(poets[index].name, poets[index].born);
}
/* output:
Wang 701
Dante 1265
Blake 1757
cummings 1894
Angelou 1928
*/
```

查询数组

`Array` 和 `Vector` 类的四种方法 `concat()`、`join()`、`slice()` 和 `toString()` 均可用于查询数组的信息，而不修改数组。`concat()` 和 `slice()` 方法返回新数组；而 `join()` 和 `toString()` 方法返回字符串。`concat()` 方法将新数组和元素列表作为参数，并将其与现有数组结合起来创建新数组。`slice()` 方法有两个名为 `startIndex` 和 `endIndex` 的参数，并返回一个新数组，新数组中包含从现有数组“分离”的元素副本。分离从 `startIndex` 处的元素开始，到 `endIndex` 处的前一个元素结束。值得强调的是，`endIndex` 处的元素不包括在返回值中。

以下示例通过 `concat()` 和 `slice()` 方法，使用其它数组的元素创建一个新数组：

```

var array1:Array = ["alpha", "beta"];
var array2:Array = array1.concat("gamma", "delta");
trace(array2); // output: alpha,beta,gamma,delta

var array3:Array = array1.concat(array2);
trace(array3); // output: alpha,beta,alpha,beta,gamma,delta

var array4:Array = array3.slice(2,5);
trace(array4); // output: alpha,beta,gamma

```

可以使用 `join()` 和 `toString()` 方法查询数组，并将其内容作为字符串返回。如果 `join()` 方法没有使用参数，则这两个方法的行为相同，它们都返回一个字符串，其中包含数组中所有元素的逗号分隔列表。与 `toString()` 方法不同，`join()` 方法接受名为 `delimiter` 的参数；可以使用此参数，选择要用作返回字符串中各个元素之间分隔符的符号。

下面的示例创建名为 `ivers` 的 `Array`，并调用 `join()` 和 `toString()` 以便采用字符串形式返回该 `Array` 中的值。`toString()` 方法用于返回以逗号分隔的值 (`iverCSV`)；而 `join()` 方法用于返回以 + 字符分隔的值。

```

var ivers:Array = ["Nile", "Amazon", "Yangtze", "Mississippi"];
var riverCSV:String = ivers.toString();
trace(riverCSV); // output: Nile,Amazon,Yangtze,Mississippi
var riverPSV:String = ivers.join("+");
trace(riverPSV); // output: Nile+Amazon+Yangtze+Mississippi

```

对于 `join()` 方法，应注意的一个问题是，无论为主数组元素指定的分隔符是什么，为嵌套 `Array` 或 `Vector` 实例返回的值始终以逗号作为分隔符，如下面的示例所示：

```

var nested:Array = ["b","c","d"];
var letters:Array = ["a",nested,"e"];
var joined:String = letters.join("+");
trace(joined); // output: a+b,c,d+e

```

关联数组

关联数组有时也称为“哈希”或“映射”，它使用“键”而非数字索引来组织存储的值。关联数组中的每个键都是用于访问一个存储值的唯一字符串。关联数组为 `Object` 类的实例，也就是说每个键都与一个属性名称对应。关联数组是键和值对的无序集合。在代码中，不应期望关联数组的键按特定的顺序排列。

ActionScript 3.0 中还引入了名为“字典”的高级关联数组类型。字典是 `flash.utils` 包中的 `Dictionary` 类的实例，使用的键可以为任意数据类型。换言之，字典的键不局限于 `String` 类型的值。

具有字符串键的关联数组

在 ActionScript 3.0 中有两种创建关联数组的方式。第一种方式是使用 `Object` 实例。使用 `Object` 实例，您可以通过对象文本来初始化数组。`Object` 类的实例（也称为“通用对象”）在功能上等同于关联数组。通用对象的每个属性名称都用作键，提供对存储的值的访问。

下面的示例创建一个名为 `monitorInfo` 的关联数组，并使用对象文本初始化具有两个键和值对的数组：

```

var monitorInfo:Object = {type:"Flat Panel", resolution:"1600 x 1200"};
trace(monitorInfo["type"], monitorInfo["resolution"]);
// output: Flat Panel 1600 x 1200

```

如果在声明数组时不需要初始化，可以使用 `Object` 构造函数创建数组，如下所示：

```
var monitorInfo:Object = new Object();
```

使用对象文本或 `Object` 类构造函数创建数组后，可以使用数组访问 `([])` 运算符或点运算符 `(.)` 向数组添加新值。`).` 以下示例将两个新值添加到 `monitorArray` 中：

```
monitorInfo["aspect ratio"] = "16:10"; // bad form, do not use spaces
monitorInfo.colors = "16.7 million";
trace(monitorInfo["aspect ratio"], monitorInfo.colors);
// output: 16:10 16.7 million
```

请注意，名为 `aspect ratio` 的键包含空格字符。可以使用数组访问 (`[]`) 运算符，但是如果尝试使用点运算符，则会生成错误。不建议在键名称中使用空格。

创建关联数组的第二种方式是使用 `Array` 构造函数（或任何动态类的构造函数），然后使用数组访问 (`[]`) 运算符或点运算符 (`.`) 将键和值对添加到数组中。如果将关联数组声明为数组类型，则将无法使用对象文本初始化该数组。以下示例使用 `Array` 构造函数创建一个名为 `monitorInfo` 的关联数组，并添加一个名为 `type` 的键和一个名为 `resolution` 的键以及它们的值：

```
var monitorInfo:Array = new Array();
monitorInfo["type"] = "Flat Panel";
monitorInfo["resolution"] = "1600 x 1200";
trace(monitorInfo["type"], monitorInfo["resolution"]);
// output: Flat Panel 1600 x 1200
```

使用 `Array` 构造函数创建关联数组没有什么优势。即使使用 `Array` 构造函数或 `Array` 数据类型，也不能将 `Array` 类的 `Array.length` 属性或任何方法用于关联数组。最好将 `Array` 构造函数用于创建索引数组。

具有对象键的关联数组（词典）

可以使用 `Dictionary` 类创建使用对象而非字符串作为键的关联数组。这样的数组有时候也称作字典、哈希或映射。例如，考虑这样一个应用程序，它可根据 `Sprite` 对象与特定容器的关联确定 `Sprite` 对象的位置。可以使用 `Dictionary` 对象，将每个 `Sprite` 对象映射到一个容器。

以下代码创建三个用作 `Dictionary` 对象的键的 `Sprite` 对象实例。它为每个键分配了值 `GroupA` 或 `GroupB`。值可以是任意数据类型，但在此示例中，`GroupA` 和 `GroupB` 均为 `Object` 类的实例。然后，可以使用数组访问 (`[]`) 运算符访问与每个键关联的值，如下面的代码所示：

```
import flash.display.Sprite;
import flash.utils.Dictionary;

var groupMap:Dictionary = new Dictionary();

// objects to use as keys
var spr1:Sprite = new Sprite();
var spr2:Sprite = new Sprite();
var spr3:Sprite = new Sprite();

// objects to use as values
var groupA:Object = new Object();
var groupB:Object = new Object();

// Create new key-value pairs in dictionary.
groupMap[spr1] = groupA;
groupMap[spr2] = groupB;
groupMap[spr3] = groupB;

if (groupMap[spr1] == groupA)
{
    trace("spr1 is in groupA");
}
if (groupMap[spr2] == groupB)
{
    trace("spr2 is in groupB");
}
if (groupMap[spr3] == groupB)
{
    trace("spr3 is in groupB");
}
```

使用对象键循环访问

您可以使用 `for..in` 循环或 `for each..in` 循环来遍历 `Dictionary` 对象的内容。`for..in` 循环用于基于键进行遍历；而 `for each..in` 循环用于基于与每个键关联的值进行遍历。

使用 `for..in` 循环可以直接访问 `Dictionary` 对象的对象键。还可以使用数组访问 (`[]`) 运算符访问 `Dictionary` 对象的值。下面的代码使用前面的 `groupMap` 字典示例来说明如何使用 `for..in` 循环来遍历 `Dictionary` 对象：

```
for (var key:Object in groupMap)
{
    trace(key, groupMap[key]);
}
/* output:
[object Sprite] [object Object]
[object Sprite] [object Object]
[object Sprite] [object Object]
*/
```

使用 `for each..in` 循环可以直接访问 `Dictionary` 对象的值。下面的代码也使用 `groupMap` 字典来说明如何使用 `for each..in` 循环来遍历 `Dictionary` 对象：

```
for each (var item:Object in groupMap)
{
    trace(item);
}
/* output:
[object Object]
[object Object]
[object Object]
*/
```

对象键和内存管理

Adobe® Flash® Player 和 Adobe® AIR™ 使用垃圾回收系统来恢复不再使用的内存。当对象不具有指向它的引用时，即可对其进行垃圾回收，并会在下次执行垃圾回收系统时恢复内存。例如，下面的代码创建一个新对象，并将对此对象的引用分配给变量 myObject：

```
var myObject:Object = new Object();
```

只要有对此对象的引用，垃圾回收系统就不会恢复此对象占用的内存。如果更改 myObject 的值以使其指向其它对象或将其设置为值 null，并且没有对原始对象的其它引用，则可以对原始对象占用的内存进行垃圾回收。

如果将 myObject 用作 Dictionary 对象中的键，则会创建对原始对象的另一个引用。例如，下面的代码创建两个对象引用（myObject 变量和 myMap 对象中的键）：

```
import flash.utils.Dictionary;
```

```
var myObject:Object = new Object();
var myMap:Dictionary = new Dictionary();
myMap[myObject] = "foo";
```

若要使 myObject 引用的对象能够进行垃圾回收，您必须删除对它的所有引用。在此情况下，必须更改 myObject 的值并从 myMap 中删除 myObject 键，如以下代码所示：

```
myObject = null;
delete myMap[myObject];
```

或者，可以使用 Dictionary 构造函数的 useWeakReference 参数，以使所有字典键均成为“弱引用”。垃圾回收系统忽略弱引用，也就是说只具有弱引用的对象可以进行垃圾回收。例如，在下面的代码中，您不需要从 myMap 中删除 myObject 键就可以使该对象能够进行垃圾回收：

```
import flash.utils.Dictionary;

var myObject:Object = new Object();
var myMap:Dictionary = new Dictionary(true);
myMap[myObject] = "foo";
myObject = null; // Make object eligible for garbage collection.
```

多维数组

多维数组将其它数组作为其元素。例如，考虑一个任务列表，它存储为有索引的字符串数组：

```
var tasks:Array = ["wash dishes", "take out trash"];
```

如果要将一周中每天的任务存储为一个单独的列表，可以创建一个多维数组，一周中的每天使用一个元素。每个元素包含一个与 tasks 数组类似的索引数组，而该索引数组存储任务列表。在多维数组中，可以使用任意组合的索引数组和关联数组。以下部分中的示例使用了两个索引数组或由索引数组组成的关联数组。练习的时候，您也许需要采用其它组合。

两个索引数组

使用两个索引数组时，可以将结果呈现为表或电子表格。第一个数组的元素表示表的行，第二个数组的元素表示表的列。

例如，以下多维数组使用两个索引数组跟踪一周中每一天的任务列表。第一个数组 masterTaskList 是使用 Array 类构造函数创建的。此数组中的各个元素分别表示一周中的各天，其中索引 0 表示星期一，索引 6 表示星期日。可将这些元素当成是表的行。可通过为 masterTaskList 数组中创建的七个元素中的每个元素分配数组文本来创建每一天的任务列表。这些数组文本表示表的列。

```
var masterTaskList:Array = new Array();
masterTaskList[0] = ["wash dishes", "take out trash"];
masterTaskList[1] = ["wash dishes", "pay bills"];
masterTaskList[2] = ["wash dishes", "dentist", "wash dog"];
masterTaskList[3] = ["wash dishes"];
masterTaskList[4] = ["wash dishes", "clean house"];
masterTaskList[5] = ["wash dishes", "wash car", "pay rent"];
masterTaskList[6] = ["mow lawn", "fix chair"];
```

可以使用数组访问`([])`运算符访问任意任务列表中的各个项。第一组括号表示一周的某一天，第二组括号表示这一天的任务列表。例如，若要检索星期三的列表中的第二项任务，请首先使用表示星期三的索引 2，然后使用表示列表中的第二项任务的索引 1。

```
trace(masterTaskList[2][1]); // output: dentist
```

若要检索星期日的列表中的第一项，请使用表示星期日的索引 6 和表示列表中的第一项任务的索引 0。

```
trace(masterTaskList[6][0]); // output: mow lawn
```

具有索引数组的关联数组

要使单个数组的访问更加方便，可以使用关联数组表示一周的各天并使用索引数组表示任务列表。通过使用关联数组可以在引用一周中特定的一天时使用点语法，但要访问关联数组的每个元素还需额外进行运行时处理。以下示例使用关联数组作为任务列表的基础，并使用键和值对来表示一周中的每一天：

```
var masterTaskList:Object = new Object();
masterTaskList["Monday"] = ["wash dishes", "take out trash"];
masterTaskList["Tuesday"] = ["wash dishes", "pay bills"];
masterTaskList["Wednesday"] = ["wash dishes", "dentist", "wash dog"];
masterTaskList["Thursday"] = ["wash dishes"];
masterTaskList["Friday"] = ["wash dishes", "clean house"];
masterTaskList["Saturday"] = ["wash dishes", "wash car", "pay rent"];
masterTaskList["Sunday"] = ["mow lawn", "fix chair"];
```

点语法通过避免使用多组括号改善了代码的可读性。

```
trace(masterTaskList.Wednesday[1]); // output: dentist
trace(masterTaskList.Sunday[0]); // output: mow lawn
```

可以使用`for..in`循环来遍历任务，但访问与每个键关联的值时必须使用数组访问`([])`运算符，而不是点语法。由于`masterTaskList`为关联数组，因而不一定会按照您所期望的顺序检索元素，如以下示例所示：

```
for (var day:String in masterTaskList)
{
    trace(day + ": " + masterTaskList[day])
}
/* output:
Sunday: mow lawn,fix chair
Wednesday: wash dishes,dentist,wash dog
Friday: wash dishes,clean house
Thursday: wash dishes
Monday: wash dishes,take out trash
Saturday: wash dishes,wash car,pay rent
Tuesday: wash dishes,pay bills
*/
```

克隆数组

Array 类不具有复制数组的内置方法。通过调用不带参数的 `concat()` 或 `slice()` 方法，可以创建数组的“浅副本”。在浅副本中，如果原始数组具有对象元素，则仅复制指向对象的引用而非对象本身。与原始数组一样，副本也指向相同的对象。对对象所做的任何更改都会在两个数组中反映出来。

在“深副本”中，还将复制原始数组中的所有对象，从而使新数组和原始数组指向不同的对象。深度复制需要多行代码，通常需要创建函数。可以将此类函数作为全局实用程序函数或 **Array** 子类的方法来进行创建。

以下示例定义一个名为 `clone()` 的函数以执行深度复制。其算法采用了一般的 Java 编程技巧。此函数创建深副本的方法是：将数组序列化为 **ByteArray** 类的实例，然后将此数组读回到新数组中。此函数接受对象，因此既可以将此函数用于索引数组，又可以将其用于关联数组，如以下代码所示：

```
import flash.utils.ByteArray;

function clone(source:Object):*
{
    var myBA:ByteArray = new ByteArray();
    myBA.writeObject(source);
    myBA.position = 0;
    return(myBA.readObject());
}
```

高级主题

扩展 **Array** 类

Array 类是少数不是最终类的核心类之一，也就是说您可以创建自己的 **Array** 子类。本部分提供了创建 **Array** 子类的示例，并讨论了在创建子类过程中会出现的一些问题。

如前所述，ActionScript 中的数组是不能指定数据类型的，但您可以创建只接受具有指定数据类型的元素的 **Array** 子类。以下部分中的示例定义名为 **TypedArray** 的 **Array** 子类，该子类中的元素限定为具有第一个参数中指定的数据类型的值。这里的 **TypedArray** 类仅用于说明如何扩展 **Array** 类，并不一定适用于生产，这有以下若干原因。第一，类型检查是在运行时而非编译时进行的。第二，当 **TypedArray** 方法遇到不匹配时，将忽略不匹配并且不会引发异常；当然，修改此方法使其引发异常也是很简单的。第三，此类无法防止使用数组访问运算符将任一类型的值插入到数组中。第四，其编码风格倾向于简洁而非性能优化。

注：可以使用此处介绍的方法创建指定类型的数组。但是，更好的方法是使用 **Vector** 对象。**Vector** 实例是真正的指定类型的数组，在性能和其它一些方面比 **Array** 类或任何子类都好。本文所进行的讨论旨在演示如何创建 **Array** 子类。

声明子类

可以使用 `extends` 关键字来指示类为 **Array** 的子类。与 **Array** 类一样，**Array** 的子类应使用 `dynamic` 属性。否则，子类将无法正常发挥作用。

以下代码显示 **TypedArray** 类的定义，该类包含一个保存数据类型的常量、一个构造函数方法和四个能够将元素添加到数组的方法。此示例省略了各方法的代码，但在以后的部分中将列出这些代码并详加说明：

```

public dynamic class TypedArray extends Array
{
    private const dataType:Class;

    public function TypedArray(...args) {}

    AS3 override function concat(...args):Array {}

    AS3 override function push(...args):uint {}

    AS3 override function splice(...args) {}

    AS3 override function unshift(...args):uint {}
}

```

由于本示例中假定将编译器选项 `-as3` 设置为 `true`, 而将编译器选项 `-es` 设置为 `false`, 因而这四个被覆盖的方法均使用 AS3 命名空间而非 `public` 属性。这些是 Adobe Flex Builder 3 和 Adobe® Flash® CS4 Professional 的默认设置。有关详细信息, 请参阅第 100 页的“[AS3 命名空间](#)”。

 如果您是倾向于使用原型继承的高级开发人员, 您可能会在以下两个方面对 `TypedArray` 类进行较小的改动, 以使其在编译器选项 `-es` 设置为 `true` 的情况下进行编译。一方面, 删除出现的所有 `override` 属性, 并使用 `public` 属性替换 AS3 命名空间。另一方面, 使用 `Array.prototype` 替换出现的所有四处 `super`。

TypedArray 构造函数

由于此子类构造函数必须接受一列任意长度的参数, 从而造成了一种有趣的挑战。该挑战就是如何将参数传递给超类构造函数以创建数组。如果将一列参数作为数组进行传递, 则超类构造函数会将其视为 `Array` 类型的一个参数, 并且生成的数组长度始终只有 1 个元素。传递参数列表的传统处理方式是使用 `Function.apply()` 方法, 此方法将参数数组作为第二个参数, 但在执行函数时将其转换为一列参数。遗憾的是, `Function.apply()` 方法不能和构造函数一起使用。

剩下的唯一方法是在 `TypedArray` 构造函数中重新创建 `Array` 构造函数的逻辑。以下代码说明在 `Array` 类构造函数中使用的算法, 您可以在 `Array` 子类构造函数中重复使用此算法:

```

public dynamic class Array
{
    public function Array(...args)
    {
        var n:uint = args.length;
        if (n == 1 && (args[0] is Number))
        {
            var dlen:Number = args[0];
            var ulen:uint = dlen;
            if (ulen != dlen)
            {
                throw new RangeError("Array index is not a 32-bit unsigned integer (" + dlen + ")");
            }
            length = ulen;
        }
        else
        {
            length = n;
            for (var i:int=0; i < n; i++)
            {
                this[i] = args[i];
            }
        }
    }
}

```

TypedArray 构造函数与 **Array** 构造函数中的大部分代码都相同，只在四个地方对代码进行了改动。其一，参数列表中新增了一个必需的 **Class** 类型参数，使用此参数可以指定数组的数据类型。其二，将传递给构造函数的数据类型分配给 **dataType** 变量。其三，在 **else** 语句中，在 **for** 循环之后为 **length** 属性赋值，以使 **length** 只包括相应类型的参数。其四，**for** 循环的主体使用 **push()** 方法的被覆盖版本，以便仅将正确数据类型的参数添加到数组中。以下示例显示 **TypedArray** 构造函数：

```
public dynamic class TypedArray extends Array
{
    private var dataType:Class;
    public function TypedArray(typeParam:Class, ...args)
    {
        dataType = typeParam;
        var n:uint = args.length
        if (n == 1 && (args[0] is Number))
        {
            var dlen:Number = args[0];
            var ulen:uint = dlen
            if (ulen != dlen)
            {
                throw new RangeError("Array index is not a 32-bit unsigned integer (" + dlen + ")")
            }
            length = ulen;
        }
        else
        {
            for (var i:int=0; i < n; i++)
            {
                // type check done in push()
                this.push(args[i])
            }
            length = this.length;
        }
    }
}
```

TypedArray 覆盖的方法

TypedArray 类覆盖前述四种能够将元素添加到数组的方法。在每种情况下，被覆盖方法均添加类型检查，这种检查可以防止添加不正确数据类型的元素。然后，每种方法均调用其自身的超类版本。

push() 方法使用 **for..in** 循环来遍历参数列表，并对每个参数执行类型检查。可以使用 **splice()** 方法，从 **args** 数组中删除所有不是正确类型的参数。在 **for..in** 循环结束后，**args** 数组仅包含 **dataType** 类型的值。然后对更新后的 **args** 数组调用 **push()** 的超类版本，如以下代码所示：

```
AS3 override function push(...args):uint
{
    for (var i:* in args)
    {
        if (!(args[i] is dataType))
        {
            args.splice(i,1);
        }
    }
    return (super.push.apply(this, args));
}
```

concat() 方法创建一个名为 **passArgs** 的临时 **TypedArray** 来存储通过类型检查的参数。这样，便可以重复使用 **push()** 方法中的类型检查代码。**for..in** 循环用于遍历 **args** 数组，并为每个参数调用 **push()**。由于将 **passArgs** 指定为类型 **TypedArray**，因而将执行 **push()** 的 **TypedArray** 版本。然后，**concat()** 方法将调用其自身的超类版本，如以下代码所示：

```
AS3 override function concat(...args):Array
{
    var passArgs:TypedArray = new TypedArray(dataType);
    for (var i:* in args)
    {
        // type check done in push()
        passArgs.push(args[i]);
    }
    return (super.concat.apply(this, passArgs));
}
```

`splice()` 方法使用任意一列参数，但前两个参数始终引用索引号和要删除的元素个数。这就是为什么被覆盖的 `splice()` 方法仅对索引位置 2 或其以后的 `args` 数组元素执行类型检查。该代码中一个有趣的地方是，`for` 循环中的 `splice()` 调用看上去似乎是递归调用，但实际上并不是递归调用，这是由于 `args` 的类型是 `Array` 而非 `TypedArray`，也就是说，`args.splice()` 调用是对此方法超类版本的调用。在 `for.in` 循环结束后，`args` 数组中将只包含索引位置 2 或其以后位置中具有正确类型的值，并且 `splice()` 会调用其自身的超类版本，如下面的代码所示：

```
AS3 override function splice(...args):*
{
    if (args.length > 2)
    {
        for (var i:int=2; i< args.length; i++)
        {
            if (!(args[i] is dataType))
            {
                args.splice(i,1);
            }
        }
    }
    return (super.splice.apply(this, args));
}
```

用于将元素添加到数组开头的 `unshift()` 方法也可以接受任意一列参数。被覆盖的 `unshift()` 方法使用的算法与 `push()` 方法使用的算法非常类似，如下面的示例代码所示：

```
AS3 override function unshift(...args):uint
{
    for (var i:* in args)
    {
        if (!(args[i] is dataType))
        {
            args.splice(i,1);
        }
    }
    return (super.unshift.apply(this, args));
}
```

示例：PlayList

此 PlayList 示例在管理歌曲列表的音乐播放列表应用程序环境中展示了使用数组的多种技巧。这些方法包括：

- 创建索引数组
- 向索引数组中添加项
- 使用不同的排序选项按照不同的属性对对象数组排序
- 将数组转换为以字符分隔的字符串

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可以在 Samples/PlayList 文件夹中找到 PlayList 应用程序文件。该应用程序包含以下文件：

文件	描述
PlayList.mxml 或 PlayList.fla	Flash 或 Flex 中的主应用程序文件（分别为 FLA 和 MXML）。
com/example/programmingas3/playlist/PlayList.as	表示歌曲列表的类。该类使用 Array 存储列表，并管理列表项的排序。
com/example/programmingas3/playlist/Song.as	表示一首歌曲信息的值对象。由 PlayList 类管理的项为 Song 实例。
com/example/programmingas3/playlist/SortProperty.as	伪枚举，其可用值代表 Song 类的属性，可以根据这些属性对 Song 对象列表排序。

PlayList 类概述

PlayList 类管理一组 Song 对象。它具有一些公共方法，可将歌曲添加到播放列表（addSong() 方法）以及对列表中的歌曲排序（sortList() 方法）。此外，它还包括一个只读存取器属性 songList，可提供对播放列表中实际歌曲集的访问。在内部，PlayList 类使用私有 Array 变量跟踪其歌曲：

```
public class PlayList
{
    private var _songs:Array;
    private var _currentSort:SortProperty = null;
    private var _needToSort:Boolean = false;
    ...
}
```

除了 PlayList 类用来跟踪其歌曲列表的 _songs Array 变量以外，还有另外两个私有变量，分别用来跟踪是否需要对列表排序（_needToSort）以及在给定时间对列表进行排序所依据的属性（_currentSort）。

跟所有对象一样，声明 Array 实例只做了创建 Array 工作的一半。在访问 Array 实例的属性或方法之前，必须在 PlayList 类的构造函数中完成对该实例的实例化。

```
public function PlayList()
{
    this._songs = new Array();
    // Set the initial sorting.
    this.sortList(SortProperty.TITLE);
}
```

该构造函数的第一行用于实例化 _songs 变量，以使其处于待用状态。此外，还会调用 sortList() 方法来设置初始排序所依据的属性。

向列表中添加歌曲

用户在应用程序中输入新歌曲后，数据条目表单上的代码将调用 PlayList 类的 addSong() 方法。

```
/***
 * Adds a song to the playlist.
 */
public function addSong(song:Song):void
{
    this._songs.push(song);
    this._needToSort = true;
}
```

在 `addSong()` 内，将调用 `_songs` 数组的 `push()` 方法，以便将传递给 `addSong()` 的 `Song` 对象添加为该数组中的新元素。不管之前应用的是哪种排序方法，使用 `push()` 方法时，都会将新元素添加到数组末尾。也就是说，调用 `push()` 方法后，歌曲列表的排序很可能不正确，因此将 `_needToSort` 变量设置为 `true`。理论上说，可以立即调用 `sortList()` 方法，而无需跟踪是否要在给定时间对列表进行排序。实际上，不需要立即对歌曲列表排序，除非要对其进行检索。通过延迟排序操作，应用程序不会执行不必要的排序，例如，在将几首歌曲添加到列表中且不需要立即对其进行检索时。

对歌曲列表排序

由于由 `PlayList` 管理的 `Song` 实例为复杂对象，因而此应用程序的用户可能要根据不同的属性（例如，歌曲名称或发行年份）来对播放列表排序。在 `PlayList` 应用程序中，对歌曲列表排序的任务由以下三部分组成：确定列表排序所依据的属性，指出按照该属性排序时应使用的排序操作，以及执行实际排序操作。

排序属性

`Song` 对象跟踪若干属性，包括歌曲名称、歌手、发行年份、文件名和用户选择的歌曲所属流派。在这些属性当中，只有前三个可用于排序。为了方便开发人员，此示例中包括 `SortProperty` 类，此类与枚举作用相同，其值表示可用于排序的属性。

```
public static const TITLE:SortProperty = new SortProperty("title");
public static const ARTIST:SortProperty = new SortProperty("artist");
public static const YEAR:SortProperty = new SortProperty("year");
```

`SortProperty` 类包含 `TITLE`、`ARTIST` 和 `YEAR` 三个常量，其中的每个常量都存储一个字符串，该字符串中包含可用于排序的相关 `Song` 类属性的实际名称。在代码的其余部分，只要指出排序属性，都是使用枚举成员完成的。例如，在 `PlayList` 构造函数中，最初通过调用 `sortList()` 方法对列表排序，如下所示：

```
// Set the initial sorting.
this.sortList(SortProperty.TITLE);
```

由于将排序属性指定为 `SortProperty.TITLE`，因而将根据歌曲名称对歌曲排序。

依据属性排序和指定排序选项

对歌曲列表排序的实际工作是由 `PlayList` 类在 `sortList()` 方法中进行的，如下所示：

```

/**
 * Sorts the list of songs according to the specified property.
 */
public function sortList(sortProperty:SortProperty):void
{
    ...
    var sortOptions:uint;
    switch (sortProperty)
    {
        case SortProperty.TITLE:
            sortOptions = Array.CASEINSENSITIVE;
            break;
        case SortProperty.ARTIST:
            sortOptions = Array.CASEINSENSITIVE;
            break;
        case SortProperty.YEAR:
            sortOptions = Array.NUMERIC;
            break;
    }

    // Perform the actual sorting of the data.
    this._songs.sortOn(sortProperty.propertyName, sortOptions);

    // Save the current sort property.
    this._currentSort = sortProperty;

    // Record that the list is sorted.
    this._needToSort = false;
}

```

如果根据歌名或歌手排序，则应按照字母顺序排序；若要根据年份排序，则按照数字顺序排序最为合理。`switch` 语句用于根据 `sortProperty` 参数中指定的值定义合适的排序选项，此选项将存储在 `sortOptions` 变量中。这里，再次使用命名的枚举成员而非硬编码值来区分不同属性。

确定排序属性和排序选项之后，将通过调用 `sortOn()` 方法并将上述两个值作为参数传递来完成对 `_songs` 数组的排序。对歌曲列表进行排序之后，记录当前排序属性。

将数组元素组合为以字符分隔的字符串

在本示例中，数组除了用于在 `PlayList` 类中维护歌曲列表以外，还用于在 `Song` 类中帮助管理指定歌曲所属的流派的列表。请考虑 `Song` 类定义中的以下代码片断：

```

private var _genres:String;

public function Song(title:String, artist:String, year:uint, filename:String, genres:Array)
{
    ...
    // Genres are passed in as an array
    // but stored as a semicolon-separated string.
    this._genres = genres.join(";");
}

```

创建新 `Song` 实例时，会将用于指定歌曲所属流派的 `genres` 参数定义为 `Array` 实例。这有助于将多个流派组合为一个可以传递给构造函数的变量。但在内部，`Song` 类在 `_genres` 私有变量中以分号分隔的 `String` 实例形式来保存流派。通过调用 `join()` 方法（使用文本字符串 ";" 作为指定分隔符），`Array` 参数将转换为以分号分隔的字符串。

通过使用相同的标记，`genres` 存取器可将流派作为 `Array` 进行设置或检索：

```
public function get genres():Array
{
    // Genres are stored as a semicolon-separated String,
    // so they need to be transformed into an Array to pass them back out.
    return this._genres.split(";");
}
public function set genres(value:Array):void
{
    // Genres are passed in as an array,
    // but stored as a semicolon-separated string.
    this._genres = value.join(";");
}
```

`genresset` 存取器的行为与构造函数完全相同；它接受 `Array` 并调用 `join()` 方法以将其转换为以分号分隔的 `String`。`get` 存取器执行相反的操作：调用 `_genres` 变量的 `split()` 方法，使用指定分隔符（采用与前面相同的文本字符串值`";"`）将 `String` 拆分为值数组。

第 9 章：处理错误

“处理”错误是指在应用程序中构建用来响应错误或修正错误的逻辑，这些错误在编译应用程序或在运行经过编译的应用程序时产生。如果应用程序能够处理错误，则在遇到错误时，应用程序会执行“一些操作”作为响应，而不是没有任何响应并且引发该错误的进程在没有提示的情况下发生失败。正确使用错误处理有助于防止应用程序和应用程序的使用者执行其它意外行为。

不过，错误处理涵盖的内容很广，它包括对编译期间或运行时引发的许多种错误予以响应。本章重点介绍如何处理运行时错误、可能生成哪些不同类型的错误以及 ActionScript 3.0 中新的错误处理系统的优点。本章还将介绍如何为您的应用程序实现您自己的自定义错误处理策略。

错误处理基础知识

错误处理简介

运行时错误是指阻止在 Adobe® Flash® Player 或 * AIR™ 中运行 ActionScript 内容的 ActionScript 代码错误。要确保为用户平稳地运行 ActionScript 代码，您必须在应用程序中编写能够处理该错误的代码，即修正该错误，解决该问题，或者至少让用户知道发生了什么错误。此过程称为“错误处理”。

错误处理涵盖的内容很广，它包括对编译期间或运行时引发的许多种错误予以响应。通常，比较容易找出在编译时发生的错误；您必须修正这些错误才能完成 SWF 文件的创建过程。本章不讨论编译时错误；有关编写不包含编译时错误的代码的详细信息，请参阅第 31 页的“[ActionScript 语言及其语法](#)”以及第 75 页的“[ActionScript 中面向对象的编程](#)”。本章重点介绍运行时错误。

运行时错误可能更难于检测，因为必须实际运行错误代码才会发生这些错误。如果程序片断包含几个代码分支（如 if..then..else 语句），您必须通过实际用户可能使用的所有输入值来测试每种可能的情况，以确保代码不包含任何错误。

运行时错误可以分为以下两类：“程序错误”是指 ActionScript 代码中的错误，如为方法参数指定了错误的数据类型；“逻辑错误”是指程序的逻辑（数据检查和值处理）错误，如在银行业应用程序中使用错误的公式来计算利率。同样，通过事先仔细地测试应用程序，通常可以检测到并纠正这两种类型的错误。

理想情况下，您希望在将应用程序发布到最终用户之前找出并消除其中的所有错误。但是，并非所有错误都是可以预见或避免的。例如，假设 ActionScript 应用程序从特定网站加载信息，而您无法控制该网站。如果该网站在某一时刻不可用，则依赖于该外部数据的应用程序部分将无法正确运行。错误处理的最重要方面涉及为这些未知情况做好准备并妥善处理它们，以使用户能够继续使用应用程序，或者至少收到一条友好的错误消息以解释无法工作的原因。

ActionScript 使用以下两种方式来表示运行时错误：

- 错误类：很多错误都具有一个关联的错误类。当发生错误时，Flash Player 或 Adobe AIR 将创建与该错误关联的特定错误类的实例。代码可以使用该错误对象中包含的信息对错误进行相应的响应。
- 错误事件：有时，当 Flash Player 或 Adobe AIR 正常触发事件时，也会发生错误。在这些情况下，Flash Player 和 Adobe AIR 触发的是错误事件。与其它事件一样，每个错误事件都有一个关联的类，Flash Player 和 Adobe AIR 会将该类的实例传递给订阅了该错误事件的方法。

要确定特定方法能否触发错误或错误事件，请参阅《[ActionScript 3.0 语言和组件参考](#)》中该方法的条目。

常见错误处理任务

您需要对代码执行的与错误处理相关的常见任务：

- 编写代码以处理错误

- 测试、捕获以及重新引发错误
- 定义您自己的错误类
- 响应错误和状态事件

重要概念和术语

以下参考列表包含将会在本章中遇到的重要术语：

- 异步：不提供即时结果的程序命令（如方法调用），而以事件形式提供结果（或错误）。
- 捕获：如果发生了异常（运行时错误）并且代码注意到该异常，则认为该代码“捕获”了异常。捕获异常后，Flash Player 和 Adobe AIR 将停止通知其它 ActionScript 代码发生了异常。
- 调试版：一种特殊的 Flash Player 或 Adobe AIR (ADL) 版本，其中包含用于通知用户发生了运行时错误的代码。在 Flash Player 或 Adobe AIR 标准版（大多数用户使用的版本）中，将忽略 ActionScript 代码未处理的错误。在调试版（Adobe Flash CS4 Professional 和 Adobe Flex 附带提供）中，当发生未处理的错误时，将显示一条警告消息。
- 异常：在程序运行时发生的错误，并且运行时环境（即 Flash Player 或 Adobe AIR）无法自行解决该问题。
- 重新引发：当代码捕获异常时，Flash Player 和 Adobe AIR 不再通知其它对象发生了异常。如果通知其它对象发生了异常非常重要，则代码必须“重新引发”异常以再次启动通知过程。
- 同步：提供即时结果或立即引发错误的程序命令（如方法调用），这意味着可以在相同的代码块内使用响应。
- 引发：通知 Flash Player 或 Adobe AIR 发生了错误（因而通知了其它对象和 ActionScript 代码）的操作称为“引发”错误。

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。实质上本章中的代码清单包括适当的 trace() 函数调用。要测试本章中的代码清单，请执行以下操作：

- 1 创建一个空的 Flash 文档。
- 2 在时间轴上选择一个关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 通过“控制”>“测试影片”运行程序。

您将在“输出”面板中看到该代码清单的 trace() 函数的结果。

后面的某些代码清单更为复杂一些，并且是以类的形式编写的。要测试这些示例，请执行以下操作：

- 1 创建一个空的 Flash 文档并将它保存到您的计算机上。
- 2 创建一个新的 ActionScript 文件，并将它保存到 Flash 文档所在的目录中。文件名应与代码清单中的类的名称一致。例如，如果代码清单定义一个名为 ErrorTest 的类，请使用名称 ErrorTest.as 来保存 ActionScript 文件。
- 3 将代码清单复制到 ActionScript 文件中并保存该文件。
- 4 在 Flash 文档中，单击舞台或工作区的空白部分，以激活文档的“属性”检查器。
- 5 在“属性”检查器的“文档类”字段中，输入您从文本中复制的 ActionScript 类的名称。
- 6 使用“控制”>“测试影片”运行程序

您将在“输出”面板中（如果示例使用 trace() 函数）或在由示例代码创建的文本字段中看到示例的结果。

用于测试示例代码清单的这些方法在第 29 页的“[测试本章内的示例代码清单](#)”中有更详细的介绍。

错误类型

开发和运行应用程序时，您会遇到不同类型的错误和错误术语。下面列出了主要的错误类型和术语：

- 编译时错误，这类错误在代码编译期间由 ActionScript 编译器引发。当代码中的语法问题导致应用程序无法生成时即会发生编译时错误。
- 运行时错误，这类错误在应用程序编译之后运行时发生。运行时错误指在 Adobe Flash Player 或 Adobe AIR 中播放 SWF 文件时产生的错误。大多数情况下，您都可以在运行时错误发生时对其进行处理，将错误报告给用户，并采取相应的步骤让应用程序继续运行。如果是致命错误，如未能连接到远程 Web 站点或未能加载需要的数据，则可以使用错误处理让应用程序妥善地终止运行。
- 同步错误，这类错误是在调用某个函数时发生的运行时错误。例如，当试图使用某个特定方法但传递给该方法的参数无效时，Flash Player 或 Adobe AIR 就会引发异常。多数错误都是在语句执行时同步发生，并且控制流会立即传递给最适用的 catch 语句。

例如，以下代码将会引发一个运行时错误，原因是在程序试图上载文件之前没有调用 `browse()` 方法：

```
var fileRef:FileReference = new FileReference();
try
{
    fileRef.upload("http://www.yourdomain.com/fileupload.cfm");
}
catch (error:IllegalOperationError)
{
    trace(error);
    // Error #2037: Functions called in incorrect sequence, or earlier
    // call was unsuccessful.
}
```

在本例中，将同步引发一个运行时错误，原因是 Flash Player 断定在试图上载文件之前没有调用 `browse()` 方法。

有关同步错误处理的详细信息，请参阅第 155 页的“[在应用程序中处理同步错误](#)”。

- 异步错误，这类错误是在应用程序运行期间在不同点处发生的运行时错误，它们会产生相应事件并由事件侦听器捕获。在异步操作中，函数发起操作但并不等待操作完成。您可以创建错误事件侦听器来等待应用程序或用户尝试执行某个操作。如果操作失败，则使用一个事件侦听器捕获错误并响应错误事件。然后，该事件侦听器调用一个事件处理函数，以便通过一种有益的方式来响应错误事件。例如，事件处理函数可以启动一个对话框，以提示用户解决该错误。

下面以前面提到的文件上载同步错误为例。如果在文件上载开始之前成功调用了 `browse()` 方法，则 Flash Player 会调度若干个事件。例如，上载开始时，将调度 `open` 事件。文件上载操作成功完成时，将调度 `complete` 事件。由于事件处理是异步的（即不是在特定、已知或预先指定的时间发生），因此需要使用 `addEventListener()` 方法来侦听这些特定的事件，如以下代码所示：

```

var fileRef:FileReference = new FileReference();
fileRef.addEventListener(Event.SELECT, selectHandler);
fileRef.addEventListener(Event.OPEN, openHandler);
fileRef.addEventListener(Event.COMPLETE, completeHandler);
fileRef.browse();

function selectHandler(event:Event):void
{
    trace("...select...");
    var request:URLRequest = new URLRequest("http://www.yourdomain.com/fileupload.cfm");
    request.method = URLRequestMethod.POST;
    event.target.upload(request.url);
}
function openHandler(event:Event):void
{
    trace("...open...");
}
function completeHandler(event:Event):void
{
    trace("...complete...");
}

```

有关异步错误处理的详细信息，请参阅第 160 页的“[响应错误事件和状态](#)”。

- 未捕获的异常，这类错误在引发后并没有相应的逻辑（如 `catch` 语句）来响应它。应用程序引发错误后，如果在当前级别或更高级别找不到适当的 `catch` 语句或事件处理函数来处理错误，则认为该错误是未捕获的异常。

鉴于用户不一定能够自行解决错误，如果错误并未使当前 SWF 文件停止播放，按照设计，Flash Player 在运行时将忽略未捕获的异常并尝试继续播放该文件。忽略未捕获异常的过程称为“无提示失败”，它会使应用程序的调试变得非常复杂。Flash Player 的调试版对未捕获异常的响应方式是终止当前脚本并在 `trace` 语句输出中显示未捕获的异常或将错误消息写入日志文件。如果异常对象是 `Error` 类或其子类的一个实例，则将调用 `getStackTrace()` 方法，并且还会在 `trace` 语句输出或日志文件中显示堆栈跟踪信息。有关使用 Flash Player 调试版的详细信息，请参阅第 154 页的“[使用 Flash Player 和 AIR 的调试版](#)”。

ActionScript 3.0 中的错误处理

由于许多应用程序在没有构建错误处理逻辑的情况下也可以运行，因此开发人员经常将在应用程序中构建错误处理逻辑的工作放在次要地位。但如果缺少错误处理逻辑，应用程序容易出现问题而造成非正常停止，或者是用户在应用程序没有按预期工作时而感到莫名其妙。ActionScript 2.0 具有一个 `Error` 类，可用来在自定义函数中构建逻辑，以便引发具有特定消息的异常。由于错误处理对于构建用户友好的应用程序至关重要，因此 ActionScript 3.0 提供了一个扩展的体系结构用来捕获错误。

注：尽管《ActionScript 3.0 语言和组件参考》中介绍了许多方法引发的异常，但可能并未涵盖每种方法可能引发的所有异常。对于某个方法，即使方法描述中确实列出了它可能引发的某些异常，但仍可能存在由于未在方法描述中明确阐述的语法错误或其它问题而引发的异常。

ActionScript 3.0 错误处理的构成元素

ActionScript 3.0 提供了许多用来进行错误处理的工具，其中包括：

- 错误类。ActionScript 3.0 中包含丰富的 `Error` 类，从而扩展了可能产生错误对象的情形的范围。每个错误类都可以帮助应用程序处理和响应特定的错误条件，无论这些错误条件是与系统错误相关（如 `MemoryError` 条件）、与代码编写错误相关（如 `ArgumentError` 条件）、与网络和通信错误相关（如 `URIError` 条件），还是与其它情形相关。有关每个类的详细信息，请参阅第 163 页的“[比较错误类](#)”。

- 更少的无提示失败。在 Flash Player 以前的版本中，只有明确使用了 `throw` 语句，才会产生并报告错误。对于 Flash Player 9 和更高版本以及 Adobe AIR，本机 ActionScript 方法和属性会引发运行时错误，使您可以在异常发生时更加有效地处理它们，然后对每个异常单独进行响应。
- 在调试期间显示清楚的错误消息。使用 Flash Player 或 Adobe AIR 的调试版时，有问题的代码或情形会生成丰富的错误消息，帮助您轻松识别特定代码块失败的原因。这会使错误的修正工作更为高效。有关详细信息，请参阅第 154 页的“[使用 Flash Player 和 AIR 的调试版](#)”。
- 精确的错误指示可以在运行时向用户显示清楚的错误消息。在 Flash Player 的先前版本中，如果 `upload()` 调用不成功，则 `FileReference.upload()` 方法会返回布尔值 `false`，表示发生了五个可能错误中的一种。在 ActionScript 3.0 中调用 `upload()` 方法时，如果发生错误，您可以引发四种特定错误中的一种，从而有助于向最终用户显示更加精确的错误消息。
- 经过优化完善的错误处理。一些明显的错误是由许多常见原因引发的。例如，在 ActionScript 2.0 中，填充 `FileReference` 对象之前，`name` 属性的值为 `null`（因此，在使用或显示 `name` 属性之前，需要确保已经设定该值且该值不为 `null`）。在 ActionScript 3.0 中，如果在填充 `name` 属性之前试图访问该属性，Flash Player 或 AIR 将引发 `IllegalOperationError`，通知您尚未设定该值，这时您可以使用 `try..catch..finally` 块来处理该错误。有关详细信息，请参阅第 155 页的“[使用 try..catch..finally 语句](#)”。
- 没有明显的性能缺点。与以前的 ActionScript 版本相比，使用 `try..catch..finally` 块仅占用很少的额外资源或根本不占用任何额外资源。
- 允许针对特定异步错误事件构建侦听器的 `ErrorEvent` 类。有关详细信息，请参阅第 160 页的“[响应错误事件和状态](#)”。

错误处理策略

只要应用程序未遇到会导致问题的条件，则即使未在程序代码中构建错误处理逻辑，应用程序仍然可以成功运行。但是，如果您没有主动处理错误，并且应用程序确实遇到了问题，用户在应用程序失败时将无从知道原因所在。

在应用程序中进行错误处理有几种不同的方式。下面概括介绍三种主要的错误处理方式：

- 使用 `try..catch..finally` 语句。这些语句可在同步错误发生时捕获它们。可以将语句嵌套在一个层次结构中，以便在不同的代码执行级别捕获异常。有关详细信息，请参阅第 155 页的“[使用 try..catch..finally 语句](#)”。
- 创建您自己的自定义错误对象。您可以使用 `Error` 类创建您自己的自定义错误对象，以便跟踪应用程序中内置错误类型未涵盖的特定操作。然后，您可以对自定义错误对象使用 `try..catch..finally` 语句。有关详细信息，请参阅第 159 页的“[创建自定义错误类](#)”。
- 编写用以响应错误事件的事件侦听器和处理函数。使用此策略，您可以创建全局错误处理函数来处理类似的事件，而无需在 `try..catch..finally` 代码块中复制许多代码。您还可以使用此方法来捕获异步错误。有关详细信息，请参阅第 160 页的“[响应错误事件和状态](#)”。

使用 Flash Player 和 AIR 的调试版

Adobe 为开发人员提供了一个特殊版本的 Flash Player 和 Adobe AIR 来帮助进行调试。安装 Adobe Flash CS4 Professional 或 Adobe Flex Builder 3 后，您可以获得一个 Flash Player 调试版副本。在安装这些工具中的任意一个时，您还可以获得 Adobe AIR 的调试版（该版本称为 ADL），该版本也包含在 Adobe AIR SDK 中。

Flash Player 和 Adobe AIR 的调试版与发行版在错误指示方面有明显的不同。调试版显示错误类型（如一般错误、`IOError` 或 `EOFError`）、错误编号和可读错误消息。发行版则仅显示错误类型和错误编号。例如，请考虑使用以下代码：

```

try
{
    tf.text = myByteArray.readBoolean();
}
catch (error:EOFError)
{
    tf.text = error.toString();
}

```

如果 `readBoolean()` 方法在 Flash Player 的调试版中引发 `EOFError`, 则会在 `tf` 文本字段中显示以下消息: “`EOFError: 错误 #2030: 到达文件末尾。`”

同样的代码在 Flash Player 或 Adobe AIR 的发行版中则会显示以下文字: “`EOFError: 错误 #2030。`”

为了使资源和大小在发行版中尽可能地小, 因此没有提供错误消息字符串。您可以通过在文档 (《ActionScript 3.0 语言和组件参考》的附录) 中查找错误编号来找到相应的错误消息。或者, 也可以使用 Flash Player 和 AIR 调试版重现错误以查看完整的错误消息。

在应用程序中处理同步错误

最常见的错误处理是同步错误处理逻辑, 您可以在处理逻辑中将适当的语句插入代码, 以便在运行时捕获同步错误。这种错误处理可以让应用程序在功能失败时注意到发生运行时错误并从错误中恢复。同步错误捕获逻辑包括 `try..catch..finally` 语句, 从字面意义上讲, 这种方式先尝试 (`try`) 某个操作, 然后捕获 (`catch`) 来自 Flash Player 的任何错误响应, 最后 (`finally`) 执行另外的操作来处理失败的操作。

使用 `try..catch..finally` 语句

处理同步运行时错误时, 可以使用 `try..catch..finally` 语句来捕获错误。当发生运行时错误时, Flash Player 或 Adobe AIR 将引发异常, 这意味着 Flash Player 或 Adobe AIR 将暂停正常的操作而创建一个 `Error` 类型的特殊对象。`Error` 对象随后会被引发到第一个可用的 `catch` 块。

`try` 语句将有可能产生错误的语句括在一起。`catch` 语句应始终与 `try` 语句一起使用。如果在 `try` 语句块的其中某个语句中检测到错误, 则会执行附加到该 `try` 语句的 `catch` 语句。

`finally` 语句将无论 `try` 语句块中是否发生错误均要执行的语句括在一起。如果没有错误, `finally` 块中的语句将在 `try` 语句块执行完毕之后执行。如果有错误, 则首先执行相应的 `catch` 语句, 然后执行 `finally` 块中的语句。

以下代码说明了使用 `try..catch..finally` 语句的语法:

```

try
{
    // some code that could throw an error
}
catch (err:Error)
{
    // code to react to the error
}
finally
{
    // Code that runs whether or not an error was thrown. This code can clean
    // up after the error, or take steps to keep the application running.
}

```

每个 `catch` 语句识别它要处理的特定类型的异常。`catch` 语句指定的错误类只能是 `Error` 类的子类。将按顺序检查每个 `catch` 语句。只执行与所引发错误的类型相匹配的第一个 `catch` 语句。换言之, 如果首先检查更高一级的 `Error` 类, 然后再检查 `Error` 类的子类, 则仅会匹配更高一级的 `Error` 类。以下代码说明了这一点:

```

try
{
    throw new ArgumentError("I am an ArgumentError");
}
catch (error:Error)
{
    trace("<Error> " + error.message);
}
catch (error:ArgumentError)
{
    trace("<ArgumentError> " + error.message);
}

```

上面这段代码的输出如下：

```
<Error> I am an ArgumentError
```

为正确捕获 **ArgumentError**，需要确保先列出最具体的错误类型，然后再列出较为一般的错误类型，如以下代码所示：

```

try
{
    throw new ArgumentError("I am an ArgumentError");
}
catch (error:ArgumentError)
{
    trace("<ArgumentError> " + error.message);
}
catch (error:Error)
{
    trace("<Error> " + error.message);
}

```

Flash Player API 中有几种方法和属性，如果在执行时它们遇到错误，便会引发运行时错误。例如，**Sound** 类中的 **close()** 方法，它如果无法关闭音频流，便会引发 **IOError** 错误，如以下代码所示：

```

var mySound:Sound = new Sound();
try
{
    mySound.close();
}
catch (error:IOError)
{
    // Error #2029: This URLStream object does not have an open stream.
}

```

随着您对《ActionScript 3.0 语言和组件参考》的逐渐熟悉，您会注意到哪些方法会引发异常（详见每个方法的说明）。

throw 语句

如果 Flash Player 和 Adobe AIR 在应用程序运行时遇到错误，便会引发异常。此外，您也可以自己使用 **throw** 语句显式引发异常。如果是显式引发错误，Adobe 建议您引发 **Error** 类或其子类的实例。以下代码所展示的 **throw** 语句引发一个 **Error** 类实例 **MyErr**，并且最后调用一个函数 **myFunction()** 在引发错误之后进行响应：

```
var MyError:Error = new Error("Encountered an error with the numUsers value", 99);
var numUsers:uint = 0;
try
{
    if (numUsers == 0)
    {
        trace("numUsers equals 0");
    }
}
catch (error:uint)
{
    throw MyError; // Catch unsigned integer errors.
}
catch (error:int)
{
    throw MyError; // Catch integer errors.
}
catch (error:Number)
{
    throw MyError; // Catch number errors.
}
catch (error:*)
{
    throw MyError; // Catch any other error.
}
finally
{
    myFunction(); // Perform any necessary cleanup here.
}
```

请注意，`catch` 语句进行了排序，以便先列出最具体的数据类型。如果先列出的是用于 `Number` 数据类型的 `catch` 语句，则无论什么情况下都不会执行用于 `uint` 数据类型的 `catch` 语句，也不会执行用于 `int` 数据类型的 `catch` 语句。

注：在 Java 编程语言中，每个可以引发异常的函数都必须先声明这一点，并在附加到函数声明的 `throw` 子句中列出该函数可以引发的异常类。ActionScript 不要求您声明函数可以引发的异常。

显示简单错误消息

新的异常和错误事件模型的一个最大优点就是：它可以让您向用户告知操作失败的时间和原因。您的工作是编写用来显示消息的代码和在响应中提供选项。

以下代码使用一个简单的 `try..catch` 语句在一个文本字段中显示错误：

```
package
{
    import flash.display.Sprite;
    import flash.text.TextField;

    public class SimpleError extends Sprite
    {
        public var employee:XML =
            <EmpCode>
                <costCenter>1234</costCenter>
                <costCenter>1-234</costCenter>
            </EmpCode>;

        public function SimpleError()
        {
            try
            {
                if (employee.costCenter.length() != 1)
                {
                    throw new Error("Error, employee must have exactly one cost center assigned.");
                }
            }
            catch (error:Error)
            {
                var errorMessage:TextField = new TextField();
                errorMessage.autoSize = TextFieldAutoSize.LEFT;
                errorMessage.textColor = 0xFF0000;
                errorMessage.text = error.message;
                addChild(errorMessage);
            }
        }
    }
}
```

与 ActionScript 的先前版本相比，ActionScript 3.0 使用了更加丰富的错误类和内置编译器错误，因此可以提供有关失败原因的更多信息。这就使您可以构建更加稳定且具有更佳错误处理能力的应用程序。

重新引发错误

构建应用程序时，有时候如果无法正确处理错误，则可能需要重新引发该错误。例如，以下代码使用一个嵌套的 try..catch 块，它在嵌套的 catch 块无法处理错误时重新引发一个自定义的 ApplicationError：

```

try
{
    try
    {
        trace("<< try >>");
        throw new ArgumentError("some error which will be rethrown");
    }
    catch (error:ApplicationError)
    {
        trace("<< catch >> " + error);
        trace("<< throw >>");
        throw error;
    }
    catch (error:Error)
    {
        trace("<< Error >> " + error);
    }
}
catch (error:ApplicationError)
{
    trace("<< catch >> " + error);
}

```

上面这个代码片断的输出如下：

```

<< try >>
<< catch >> ArgumentError: some error which will be rethrown
<< throw >>
<< catch >> ArgumentError: some error which will be rethrown

```

嵌套的 try 块引发一个自定义的 `ArgumentError` 错误，该错误由后续 `catch` 块捕获。此嵌套的 `catch` 块会尝试处理错误，如果不成功，则将 `ArgumentError` 对象引发到包含此 `catch` 块的 `try..catch` 块中来进行处理。

创建自定义错误类

您可以通过扩展其中一种标准的错误类，在 ActionScript 中创建您自己的专用错误类。有多种原因需要创建您自己的错误类：

- 识别应用程序特有的错误或错误组。

例如，除了由 Flash Player 或 Adobe AIR 捕获的错误外，您可能希望对您自己的代码所引发的错误采取另外的操作。您可以创建 `Error` 类的一个子类，以便在 `try..catch` 块中跟踪新的错误数据类型。

- 为应用程序生成的错误提供特有的错误显示能力。

例如，可以创建一个以某种方式设置错误消息格式的新的 `toString()` 方法。还可以定义一个 `lookupErrorString()` 方法，该方法获取错误代码并根据用户的语言首选参数查找适当的消息。

专用的错误类必须扩展 ActionScript 的核心错误类。以下是一个扩展了 `Error` 类的专用 `AppError` 类示例：

```

public class AppError extends Error
{
    public function AppError(message:String, errorID:int)
    {
        super(message, errorID);
    }
}

```

以下是在项目中使用 `AppError` 的一个示例：

```
try
{
    throw new AppError("Encountered Custom AppError", 29);
}
catch (error:AppError)
{
    trace(error.errorID + ": " + error.message)
}
```

注: 如果要在子类中覆盖 `Error.toString()` 方法, 则需要在该方法中使用 ... 参数 (表示还有其它参数)。ActionScript 3.0 所依据的 ECMAScript 语言规范通过这种方式定义 `Error.toString()` 方法, ActionScript 3.0 通过同样的方式定义该方法以实现向后兼容。因此, 在覆盖 `Error.toString()` 方法时, 必须精确匹配各个参数。运行时无需向 `toString()` 方法传递任何参数, 因为传递的参数都会被忽略。

响应错误事件和状态

在 ActionScript 3.0 中, 对错误处理最为明显的一项改进就是支持对异步运行时错误予以响应。(有关异步错误的定义, 请参阅第 152 页的“[错误类型](#)”。)

可以创建事件侦听器和事件处理函数来响应错误事件。对于许多类来说, 它们调度错误事件的方式与调度其它事件的方式相同。例如, 一般情况下, `XMLSocket` 类实例调度三种类型的事件: `Event.CLOSE`、`Event.CONNECT` 和 `DataEvent.DATA`。但是, 当发生问题时, `XMLSocket` 类还可以调度 `IOErrorEvent.IOError` 或 `SecurityErrorEvent.SECURITY_ERROR`。有关事件侦听器和事件处理函数的详细信息, 请参阅第 207 页的“[处理事件](#)”。

错误事件分为两类:

- 扩展 `ErrorEvent` 类的错误事件

`flash.events.ErrorEvent` 类包含用于管理有关网络和通信操作的运行时错误的属性和方法。`AsyncErrorEvent`、`IOErrorEvent` 和 `SecurityErrorEvent` 类扩展了 `ErrorEvent` 类。如果使用的是 Flash Player 或 Adobe AIR 的调试版, 则会出现一个对话框, 向您告知播放器在运行时遇到的、没有侦听器函数的错误事件。

- 基于状态的错误事件

基于状态的错误事件与网络和通信类的 `netStatus` 和 `status` 属性有关。如果 Flash Player 或 Adobe AIR 在读写数据时遇到问题, `netStatus.info.level` 或 `status.level` 属性 (取决于使用的类对象) 的值将被设置为值 "error"。可以通过检查事件处理函数中的 `level` 属性是否包含值 "error" 来响应此错误。

处理错误事件

`ErrorEvent` 类及其子类包含用于处理 Flash Player 和 Adobe AIR 尝试读写数据时调度的错误的错误类型。

下面的示例同时使用了 `try..catch` 语句和错误事件处理函数来显示在尝试读取本地文件时检测到的任何错误。您可以在由“在此处添加您的错误处理代码”注释所指示的位置处添加更复杂的处理代码, 以便为用户提供处理选项或者自动处理错误:

```
package
{
    import flash.display.Sprite;
    import flash.errors.IOError;
    import flash.events.IOErrorEvent;
    import flash.events.TextEvent;
    import flash.media.Sound;
    import flash.media.SoundChannel;
    import flash.net.URLRequest;
    import flash.text.TextField;
    import flash.text.TextFieldAutoSize;

    public class LinkEventExample extends Sprite
    {
        private var myMP3:Sound;
        public function LinkEventExample()
        {
            myMP3 = new Sound();
            var list:TextField = new TextField();
            list.autoSize = TextFieldAutoSize.LEFT;
            list.multiline = true;
            list.htmlText = "<a href=\"event:track1.mp3\">Track 1</a><br>";
            list.htmlText += "<a href=\"event:track2.mp3\">Track 2</a><br>";
            addEventListener(TextEvent.LINK, linkHandler);
            addChild(list);
        }

        private function playMP3(mp3:String):void
        {
            try
            {
                myMP3.load(new URLRequest(mp3));
                myMP3.play();
            }
            catch (err:Error)
            {
                trace(err.message);
                // your error-handling code here
            }
            myMP3.addEventListener(IOErrorEvent.IO_ERROR, errorHandler);
        }

        private function linkHandler(linkEvent:TextEvent):void
        {
            playMP3(linkEvent.text);
            // your error-handling code here
        }

        private function errorHandler(errorEvent:IOErrorEvent):void
        {
            trace(errorEvent.text);
            // your error-handling code here
        }
    }
}
```

处理状态更改事件

对于支持 level 属性的类，Flash Player 和 Adobe AIR 会动态更改 netStatus.info.level 或 status.level 属性的值。具有 netStatus.info.level 属性的类有 NetConnection、NetStream 和 SharedObject。具有 status.level 属性的类有 HTTPStatusEvent、Camera、Microphone 和 LocalConnection。可以编写一个处理函数来响应 level 值的更改并跟踪通信错误。

以下示例使用 netStatusHandler() 函数测试 level 属性的值。如果 level 属性指示遇到错误，该代码将跟踪消息“Video stream failed”（视频流失败）。

```
package
{
    import flash.display.Sprite;
    import flash.events.NetStatusEvent;
    import flash.events.SecurityErrorEvent;
    import flash.media.Video;
    import flash.net.NetConnection;
    import flash.net.NetStream;

    public class VideoExample extends Sprite
    {
        private var videoUrl:String = "Video.flv";
        private var connection:NetConnection;
        private var stream:NetStream;

        public function VideoExample()
        {
            connection = new NetConnection();
            connection.addEventListener(NetStatusEvent.NET_STATUS, netStatusHandler);
            connection.addEventListener(SecurityErrorEvent.SECURITY_ERROR, securityErrorHandler);
            connection.connect(null);
        }

        private function netStatusHandler(event:NetStatusEvent):void
        {
            if (event.info.level == "error")
            {
                trace("Video stream failed");
            }
            else
            {
                connectStream();
            }
        }

        private function securityErrorHandler(event:SecurityErrorEvent):void
        {
            trace("securityErrorHandler: " + event);
        }

        private function connectStream():void
        {
            var stream:NetStream = new NetStream(connection);
            var video:Video = new Video();
            video.attachNetStream(stream);
            stream.play(videoUrl);
            addChild(video);
        }
    }
}
```

比较错误类

ActionScript 提供了一些预定义的 Error 类。其中的许多错误类由 Flash Player 和 Adobe AIR 使用，但您在自己的代码中也可以使用这些错误类。在 ActionScript 3.0 中，有两种主要类型的错误类：ActionScript 核心错误类和 flash.error 包错误类。核心 Error 类由 ECMAScript (ECMA-262) 第 3 版语言规范定义。flash.error 包包含为有助于 ActionScript 3.0 应用程序开发和调试工作而引入的附加类。

ECMAScript 核心错误类

ECMAScript 核心错误类包括 Error、EvalError、RangeError、ReferenceError、SyntaxError、TypeError 以及 URIError 类。其中的每个类均位于顶级命名空间中。

类名称	描述	注释
Error	Error 类可用于引发异常，它是 ECMAScript 中所定义的其它异常类 (EvalError、RangeError、ReferenceError、SyntaxError、TypeError 和 URIError) 的基类。	Error 类用作 Flash® Player 和 Adobe® AIR® 所引发的所有运行时错误的基类，并建议将其用作任何自定义错误类的基类。
EvalError	如果为 Function 类的构造函数传递了任何参数，或者用户代码调用 eval() 函数，则会引发 EvalError 异常。	在 ActionScript 3.0 中，取消了对 eval() 函数的支持，因此，如果试图使用该函数，则会引发错误。 Flash Player 的先前版本使用 eval() 函数来按照名称访问变量、属性、对象或影片剪辑。
RangeError	如果数值不在可接受的范围内，则会引发 RangeError 异常。	例如，当延迟是负数或不是有限值时，Timer 类就会引发 RangeError。试图在无效深度处添加显示对象也会引发 RangeError。
ReferenceError	如果试图对密封（非动态）对象引用未定义的属性，则会引发 ReferenceError 异常。试图访问 undefined 的属性时，ActionScript 3.0 之前的 ActionScript 编译器版本并不会引发错误。然而，ActionScript 3.0 在此情况下会引发 ReferenceError 异常。	由于访问未定义变量而引发异常表明存在潜在的错误，有助于提高软件质量。不过，如果您不习惯于对变量进行初始化，则对于 ActionScript 这种新的行为，您可能需要改变一些代码编写习惯。

类名称	描述	注释
SyntaxError	<p>如果 ActionScript 代码发生解析错误，则会引发 SyntaxError 异常。</p> <p>有关详细信息，请参阅位于以下地址的 ECMAScript (ECMA-262) 第 3 版语言规范的 15.11.6.4 节：www.ecma-international.org/publications/standards/Ecma-262.htm，以及位于以下地址的 ECMAScript for XML (E4X) 规范 (ECMA-357 第 2 版) 的 10.3.1 节：www.ecma-international.org/publications/standards/Ecma-357.htm。</p>	<p>在以下情况下会引发 SyntaxError：</p> <ul style="list-style-type: none"> 当 RegExp 类解析的是无效正则表达式时，ActionScript 会引发 SyntaxError 异常。 当 XMLDocument 类解析的是无效 XML 时，ActionScript 会引发 SyntaxError 异常。
TypeError	<p>如果操作数的实际类型与所需类型不同，则会引发 TypeError 异常。</p> <p>有关详细信息，请参阅位于以下地址的 ECMAScript 规范的 15.11.6.5 节：www.ecma-international.org/publications/standards/Ecma-262.htm，以及位于以下地址的 E4X 规范的 10.3 节：www.ecma-international.org/publications/standards/Ecma-357.htm。</p>	<p>在以下情况下会引发 TypeError：</p> <ul style="list-style-type: none"> 无法将函数的实际参数或方法强制为正式参数类型。 值已赋给变量，但无法强制为变量的类型。 is 或 instanceof 运算符右侧的内容不是有效类型。 非法使用了 super 关键字。 属性查找生成了多个绑定，因此造成该查找不明确。 对不兼容的对象调用了某种方法。例如，如果 RegExp 类中的某个方法被“转接”到通用对象上，然后调用该方法，则将引发 TypeError 异常。
URIError	<p>如果采用与某个全局 URI 处理函数的定义相矛盾的方式使用该函数，则会引发 URIError 异常。</p> <p>有关详细信息，请参阅位于以下地址的 ECMAScript 规范的 15.11.6.6 节：www.ecma-international.org/publications/standards/Ecma-262.htm。</p>	<p>在以下情况下会引发 URIError：</p> <p>为需要有效 URI 的 Flash Player API 函数（如 Socket.connect()）指定了无效的 URI。</p>

ActionScript 核心错误类

除 ECMAScript 核心错误类之外，ActionScript 还增加了几个自己的类，用于 ActionScript 特定的错误条件和错误处理功能。

由于这些类是对 ECMAScript 第 3 版语言规范的 ActionScript 语言扩展，并且可能会被收入该规范的未来版本中，因此将这些类保留在顶级，而不是放置在某个包（如 `flash.error`）中。

类名称	描述	注释
ArgumentError	ArgumentError 类表示在函数调用期间提供的参数与为该函数定义的参数不一致时发生的错误。	<p>以下是一些参数错误示例：</p> <ul style="list-style-type: none"> 向方法提供的参数过少或过多。 参数应是某个枚举值，但实际上不是。
SecurityError	如果发生安全违规且访问被拒绝时，则会引发 SecurityError 异常。	<p>以下是一些安全错误示例：</p> <ul style="list-style-type: none"> 通过安全沙箱边界进行未经授权的属性访问或方法调用。 尝试访问安全沙箱不允许的 URL。 尝试与某个端口进行套接字连接，但是必需的套接字策略文件不存在。 尝试访问用户的摄像头或麦克风，而访问此类设备的请求被用户拒绝。
VerifyError	如果遇到格式不正确或损坏的 SWF 文件，则会引发 VerifyError 异常。	当某个 SWF 文件加载另一个 SWF 文件时，父 SWF 可能会捕获由加载的 SWF 生成的 VerifyError。

flash.error 包的错误类

flash.error 包中包含的错误类是 Flash Player API 的一部分。与刚刚描述的错误类不同， flash.error 包可以与特定于 Flash Player 或 Adobe AIR 的错误事件进行通信。

类名称	描述	注释
EOFError	如果尝试读取的内容超出可用数据的末尾，则会引发 EOFError 异常。	例如，当调用 IDataInput 接口中的一个读取方法，而数据不足以满足读取请求时，将引发 EOFError。
IllegalOperationError	如果方法未实现或者实现中未涵盖当前用法，则会引发 IllegalOperationError 异常。	<p>以下是非法操作错误异常的示例：</p> <ul style="list-style-type: none"> 基类（如 DisplayObjectContainer）提供的功能比舞台可以支持的功能多。例如，如果试图在舞台上获取或设置一个遮罩（使用 stage.mask），Flash Player 和 Adobe AIR 将引发 IllegalOperationError，并显示消息“Stage 类未实现此属性或方法”。 子类继承了不需要且不想支持的方法。 在没有辅助功能支持的情况下编译 Flash Player 之后，又调用了某些辅助功能方法。 从 Flash Player 的运行时版本中调用仅用于创作的功能。 试图为放在时间轴上的对象设置名称。
IOError	如果发生某种类型的 I/O 异常，则会引发 IOError 异常。	例如，当试图对尚未连接或已断开连接的套接字进行读 / 写操作时，将引发此错误。

类名称	描述	注释
MemoryError	如果内存分配请求失败，则会引发 <code>MemoryError</code> 异常。	默认情况下，ActionScript Virtual Machine 2 不会对 ActionScript 程序可以分配的内存量强加限制。对于桌面 PC，内存分配失败的情况并不常见。当系统无法分配操作所需的内存时，将会看到这样的错误。因此，对于桌面 PC，这种异常很少见，除非分配请求非常大，如分配 3 GB 字节的请求便无法实现，因为 32 位 Microsoft® Windows® 程序只能访问 2 GB 的地址空间。
ScriptTimeoutError	如果达到了 15 秒的脚本超时间隔，则会引发 <code>ScriptTimeoutError</code> 异常。通过捕获 <code>ScriptTimeoutError</code> 异常，可以更加妥善地处理脚本超时。如果没有异常处理函数，未捕获异常的处理函数将显示一个带有错误消息的对话框。	为防止恶意开发者捕获这种异常并导致无限循环，仅能够捕获特定脚本运行过程中引发的第一个 <code>ScriptTimeoutError</code> 异常。后面的 <code>ScriptTimeoutError</code> 异常将无法被开发者的代码捕获，并会立即转到未捕获异常的处理函数。
StackOverflowError	如果脚本可用堆栈已经用尽，则会引发 <code>StackOverflowError</code> 异常。	<code>StackOverflowError</code> 异常可能表明发生了无限递归。

示例：CustomErrors 应用程序

`CustomErrors` 应用程序展示了构建应用程序时处理自定义错误的一些技巧。这些方法包括：

- 验证 XML 包
- 编写自定义错误
- 引发自定义错误
- 引发错误时通知用户

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可以在 Samples/CustomError 文件夹中找到 `CustomErrors` 应用程序的文件。该应用程序包含以下文件：

文件	描述
CustomErrors.mxml 或 CustomErrors.fla	Flash (FLA) 或 Flex (MXML) 中的主应用程序文件
com/example/programmingas3/errors/ApplicationError.as	一个类，用作 <code>FatalError</code> 类和 <code>WarningError</code> 类的基错误类。
com/example/programmingas3/errors/FatalError.as	一个类，定义该应用程序可以引发的 <code>FatalError</code> 错误。该类扩展了自定义的 <code>ApplicationError</code> 类。
com/example/programmingas3/errors/Validator.as	一个类，定义了用于验证用户提供的员工 XML 包的单个方法。
com/example/programmingas3/errors/WarningError.as	一个类，定义了该应用程序可以引发的 <code>WarningError</code> 错误。该类扩展了自定义的 <code>ApplicationError</code> 类。

CustomErrors 应用程序概述

当应用程序加载时，将在 Flex 中调用 `initApp()` 方法，或者在 Flash 中执行时间轴（非函数）代码。此代码定义将由 `Validator` 类验证的示例 XML 包。以下代码运行：

```

employeeXML =
<employee id="12345">
    <firstName>John</firstName>
    <lastName>Doe</lastName>
    <costCenter>12345</costCenter>
    <costCenter>67890</costCenter>
</employee>;
}

```

稍后，将在舞台上的 `TextArea` 组件实例中显示该 XML 包。这样就可以在尝试重新验证该 XML 包之前对其进行修改。

用户单击 `Validate` 按钮时，将调用 `validateData()` 方法。该方法使用 `Validator` 类中的 `validateEmployeeXML()` 方法来验证员工 XML 包。以下显示的是 `validateData()` 方法的代码：

```

function validateData():void
{
    try
    {
        var tempXML:XML = XML(xmlText.text);
        Validator.validateEmployeeXML(tempXML);
        status.text = "The XML was successfully validated.";
    }
    catch (error:FatalError)
    {
        showFatalError(error);
    }
    catch (error:WarningError)
    {
        showWarningError(error);
    }
    catch (error:Error)
    {
        showGenericError(error);
    }
}

```

首先，使用 `TextArea` 组件实例 `xmlText` 的内容创建一个临时的 XML 对象。接下来，调用自定义 `Validator` 类 (`com.example.programmingas3/errors/Validator.as`) 中的 `validateEmployeeXML()` 方法，并将这个临时 XML 对象作为参数传递给该方法。如果这个 XML 包是有效的，`Label` 组件实例 `status` 便会显示一条成功消息，然后应用程序退出。如果 `validateEmployeeXML()` 方法引发了一个自定义错误（即发生 `FatalError`、`WarningError` 或一般的 `Error`），则会执行相应的 `catch` 语句并调用 `showFatalError()`、`showWarningError()` 或 `showGenericError()` 方法。这几种方法均会在一个名为 `statusText` 的文本区域中显示相应的消息，以通知用户发生了特定错误。每个方法还会用具体的消息更新 `Label` 组件实例 `status`。

如以下代码所示，如果尝试验证员工 XML 包时发生致命错误，则会在一个 `statusText` 文本区域中显示错误消息，并且禁用 `xmlText` `TextArea` 组件实例和 `validateBtn` `Button` 组件实例：

```

function showFatalError(error:FatalError):void
{
    var message:String = error.message + "\n\n";
    var title:String = error.getTitle();
    statusText.text = message + " " + title + "\n\nThis application has ended.";
    this.xmlText.enabled = false;
    this.validateBtn.enabled = false;
    hideButtons();
}

```

如果发生的是警告错误而不是致命错误，则会在 `statusText` `TextArea` 实例中显示错误消息，但不会禁用 `xmlText` `TextField` 和 `Button` 组件实例。`showWarningError()` 方法会在 `statusText` 文本区域中显示自定义的错误消息。该消息还会让用户决定是否继续验证 XML 还是终止脚本。以下显示的是 `showWarningError()` 方法的代码：

```

function showWarningError(error:WarningError):void
{
    var message:String = error.message + "\n\n" + "Do you want to exit this application?";
    showButtons();
    var title:String = error.getTitle();
    statusText.text = message;
}

```

当用户单击 Yes 或 No 按钮时，将调用 closeHandler() 方法。以下显示的是 closeHandler() 方法的代码：

```

function closeHandler(event:CloseEvent):void
{
    switch (event.detail)
    {
        case yesButton:
            showFatalError(new FatalError(9999));
            break;
        case noButton:
            statusText.text = "";
            hideButtons();
            break;
    }
}

```

如果用户选择单击 Yes 来终止脚本，则会引发 FatalError，从而导致应用程序终止运行。

构建自定义验证程序

自定义的 Validator 类仅包含一个方法：validateEmployeeXML()。validateEmployeeXML() 方法使用单一参数 employee，该参数为要验证的 XML 包。validateEmployeeXML() 方法的代码如下所示：

```

public static function validateEmployeeXML(employee:XML):void
{
    // checks for the integrity of items in the XML
    if (employee.costCenter.length() < 1)
    {
        throw new FatalError(9000);
    }
    if (employee.costCenter.length() > 1)
    {
        throw new WarningError(9001);
    }
    if (employee.ssn.length() != 1)
    {
        throw new FatalError(9002);
    }
}

```

员工必须属于一个（且只能属于一个）成本中心才能通过验证。如果员工不属于任何成本中心，该方法将引发 FatalError，该异常将向上冒泡到应用程序主文件中的 validateData() 方法。如果员工属于多个成本中心，则引发 WarningError。该 XML 验证程序最后检查用户是否只定义了一个社会安全号码（XML 包中的 ssn 节点）。如果具有不止一个 ssn 节点，则引发 FatalError 错误。

您可以向 validateEmployeeXML() 方法添加其它检查，例如，确保 ssn 节点包含的是有效号码，或者员工至少定义了一个电话号码和一个电子邮件地址，并且这两个值都是有效的。您还可以对该 XML 进行修改，使每个员工具有唯一的 ID 并指定其管理者的 ID。

定义 ApplicationError 类

`ApplicationError` 类用作 `FatalError` 类和 `WarningError` 类的基类。`ApplicationError` 类扩展了 `Error` 类，并且定义了自己的自定义方法和属性，其中包括定义一个错误 ID、严重程度以及包含自定义错误代码和消息的 XML 对象。该类还定义了两个静态常量，它们用于定义每种错误类型的严重程度。

`ApplicationError` 类的构造函数方法如下所示：

```
public function ApplicationError()
{
    messages =
    <errors>
        <error code="9000">
            <! [CDATA[Employee must be assigned to a cost center.]]>
        </error>
        <error code="9001">
            <! [CDATA[Employee must be assigned to only one cost center.]]>
        </error>
        <error code="9002">
            <! [CDATA[Employee must have one and only one SSN.]]>
        </error>
        <error code="9999">
            <! [CDATA[The application has been stopped.]]>
        </error>
    </errors>;
}
```

XML 对象中的每个错误节点都包含一个唯一的数值代码和一条错误消息。使用 E4X 可以很容易地通过错误代码查找到相应的错误消息，如以下 `getMessageText()` 方法所示：

```
public function getMessageText(id:int):String
{
    var message:XMLList = messages.error.(@code == id);
    return message[0].text();
}
```

`getMessageText()` 方法仅使用一个整数参数 `id` 并返回一个字符串。该 `id` 参数就是要查找的错误的错误代码。例如，传递等于 9001 的 `id` 值将得到错误消息“Employee must be assigned to only one cost center”（只能为员工指定一个成本中心）。如果多个错误具有同一错误代码，ActionScript 将只为找到的第一个结果返回错误消息（所返回 `XMLList` 对象中的 `message[0]`）。

该类中的下一个方法 `getTitle()` 不使用任何参数，并返回一个字符串值，其中包含此特定错误的错误 ID。该值用于帮助您轻松识别在验证 XML 包期间发生的具体错误。以下显示的是 `getTitle()` 方法的代码：

```
public function getTitle():String
{
    return "Error #" + id;
}
```

`ApplicationError` 类中的最后一个方法是 `toString()`。该方法覆盖 `Error` 类中定义的函数，以便您可以自定义错误消息的显示。该方法将返回一个字符串，用于识别所发生错误的具体编号和消息。

```
public override function toString():String
{
    return "[APPLICATION ERROR #" + id + "] " + message;
}
```

定义 FatalError 类

`FatalError` 类扩展了自定义的 `ApplicationError` 类并定义三个方法：`FatalError` 构造函数、`getTitle()` 和 `toString()`。第一个方法（即 `FatalError` 构造函数）接受一个整数参数 `errorID`，并使用 `ApplicationError` 类中定义的静态常量设置错误的严重性，另外还调用 `ApplicationError` 类中的 `getMessageText()` 方法获取特定错误的错误消息。`FatalError` 构造函数如下所示：

```
public function FatalError(errorID:int)
{
    id = errorID;
    severity = ApplicationError.FATAL;
    message = getMessageText(errorID);
}
```

FatalError 类中的下一个方法 getTitle() 覆盖先前在 ApplicationError 类中定义的 getTitle() 方法，并在标题后面追加文字“--FATAL”以通知用户发生了致命错误。getTitle() 方法如下所示：

```
public override function getTitle():String
{
    return "Error #" + id + " -- FATAL";
}
```

该类中的最后一个方法 toString() 覆盖 ApplicationError 类中定义的 toString() 方法。toString() 方法如下所示：

```
public override function toString():String
{
    return "[FATAL ERROR #" + id + "] " + message;
}
```

定义 WarningError 类

WarningError 类扩展了 ApplicationError 类，它与 FatalError 类几乎完全相同，只是字符串稍微有些不同，另外它还将错误严重程度设置为 ApplicationError.WARNING 而不是 ApplicationError.FATAL，如以下代码所示：

```
public function WarningError(errorID:int)
{
    id = errorID;
    severity = ApplicationError.WARNING;
    message = super.getMessageText(errorID);
}
```

第 10 章：使用正则表达式

正则表达式描述用于查找和处理字符串中的匹配文本的模式。正则表达式类似于字符串，但是可以包含特殊代码以描述模式和重复。例如，下面的正则表达式与以字符 A 开头并且后跟一个或多个连续数字的字符串匹配：

```
/A\d+/"
```

本章介绍了构造正则表达式的基本语法。但是实际上，正则表达式可能非常复杂且具有许多细微差别。您可以从网上或者书店中找到有关正则表达式的详细资料。切记，不同的编程环境实现正则表达式的方式也不同。ActionScript 3.0 按照 ECMAScript 第 3 版语言规范 (ECMA-262) 中的定义实现正则表达式。

正则表达式基础知识

使用正则表达式简介

正则表达式描述字符模式。通常，正则表达式用于验证文本值是否符合特定模式（例如，验证用户输入的电话号码位数是否正确），或者替换与特定模式匹配的部分文本值。

正则表达式可能非常简单。例如，假设您要确认特定字符串与“ABC”是否匹配，或者要使用某些其它文本替换字符串中出现的每个“ABC”。在这种情况下，您可以使用以下正则表达式，它定义了依次包含字母 A、B 和 C 的模式：

```
/ABC/
```

请注意，正则表达式文本是使用正斜杠 (/) 字符界定的。

正则表达式模式也可能非常复杂，有时候表面上看起来晦涩难懂，例如，以下与有效电子邮件地址匹配的表达式：

```
/([0-9a-zA-Z]+[-._+&])*[0-9a-zA-Z]+@[(-0-9a-zA-Z)+[.]]+[a-zA-Z]{2,6}/
```

通常，您使用正则表达式在字符串中搜索模式以及替换字符。在这些情况下，您将创建一个正则表达式对象，并将其作为几个 String 类方法之一的参数。下列 String 类方法将正则表达式作为参数：match()、replace()、search() 和 split()。有关这些方法的详细信息，请参阅第 121 页的“[在字符串中查找模式并替换子字符串](#)”。

RegExp 类包含以下方法：test() 和 exec()。有关详细信息，请参阅第 183 页的“[对字符串使用正则表达式的方法](#)”。

常见正则表达式任务

正则表达式具有几种常见用途，本章详细介绍了这些用途：

- 创建正则表达式模式
- 在模式中使用特殊字符
- 标识多个字符的序列（如“两位数”或“介于 7 到 10 个字母之间”）
- 标识字母或数字范围中的任何字符（如“从 a 到 m 的任何字母”）
- 标识可能的字符集中的字符
- 标识子序列（模式内的片断）
- 基于模式匹配和替换文件

重要概念和术语

以下参考列表包含将会在本章中使用的重要术语：

- 转义字符：此字符指示应将后面的字符视为元字符，而不是字面字符。在正则表达式语法中，反斜杠字符 (\) 就是转义字符，因此反斜杠后跟另一个字符是一个特殊代码，而不仅仅是字符本身。
- 标志：此字符指定有关应如何使用正则表达式模式的一些选项，如是否区分大写和小写字符。
- 元字符：在正则表达式模式中具有特殊含义的字符，它与从字面意义上在模式中表示该字符相对。
- 数量表示符：一个或几个字符，指示应将模式部分重复多少次。例如，使用数量表示符来指定美国邮政编码应包含 5 个或 9 个数字。
- 正则表达式：用于定义字符模式的程序语句，它可用来确认其它字符串是否与该模式匹配，或者替换字符串的一部分。

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。由于本章中的代码清单主要由正则表达式模式组成，因此测试示例涉及以下几个步骤：

- 1 创建一个新的 Flash 文档。
- 2 选择一个关键帧并打开“动作”面板。
- 3 创建一个 RegExp（正则表达式）变量，例如：

```
var pattern:RegExp = /ABC/;
```

- 4 复制示例的模式并将其作为 RegExp 变量的值进行分配。例如，在上一行代码中，模式是等号右侧的代码的一部分，不包含分号 (/ABC/)。
- 5 创建一个或多个包含适合于测试正则表达式的字符串的 String 变量。例如，如果要创建正则表达式来测试有效的电子邮件地址，则创建几个包含有效和无效电子邮件地址的 String 变量：

```
var goodEmail:String = "bob@example.com";
var badEmail:String = "5@$2.99";
```

- 6 添加几行测试 String 变量的代码，以确定它们是否与正则表达式模式匹配。这些将是您希望通过使用 trace() 函数或通过写入舞台的文本字段而输出到屏幕上的值。

```
trace(goodEmail, " is valid:", pattern.test(goodEmail));
trace(badEmail, " is valid:", pattern.test(badEmail));
```

例如，假设 pattern 为有效的电子邮件地址定义了正则表达式模式，前面几行代码将此文本写入“输出”面板：

```
bob@example.com is valid: true
5@$2.99 is valid: false
```

有关通过将值写入舞台上的文本字段实例或使用 trace() 函数将值输出到“输出”面板来测试这些值的详细信息，请参阅第 29 页的“[测试本章内的示例代码清单](#)”。

正则表达式语法

本节介绍了 ActionScript 正则表达式语法的全部元素。正如您所看到的一样，正则表达式可能非常复杂且具有许多细微差别。您可以从网上或者书店中找到有关正则表达式的详细资料。切记，不同的编程环境实现正则表达式的方式也不同。

ActionScript 3.0 按照 ECMAScript 第 3 版语言规范 (ECMA-262) 中的定义实现正则表达式。

通常，您要使用的正则表达式是与比较复杂的模式匹配，而不是与简单的字符串匹配。例如，下面的正则表达式定义了由字母 A、B 和 C 依次排列且后跟数字的模式：

```
/ABC\d/
```

\d 代码表示“任意数字”。反斜杠 (\) 字符称为转义字符，它与后面的字符（在本例中为字母 d）配合使用，在正则表达式中具有特殊含义。本章将介绍这些转义字符序列以及其它一些正则表达式语法特征。

下面的正则表达式定义了由字母 ABC 后跟任意数目的数字组成的模式（注意星号）：

```
/ABC\d*/
```

星号字符 (*) 是“元字符”。元字符是在正则表达式中具有特殊含义的字符。星号是一种称为“数量表示符”的特定类型的元字符，用于定义某个字符或一组字符重复的次数。有关详细信息，请参阅第 177 页的“[数量表示符](#)”。

除了它的模式外，正则表达式还可以包含标志，用于指定正则表达式的匹配方式。例如，下面的正则表达式使用 i 标志指定正则表达式在匹配字符串中忽略大小写：

```
/ABC\d*/i
```

有关详细信息，请参阅第 180 页的“[标志和属性](#)”。

您可以通过以下 String 类方法使用正则表达式：match()、replace() 和 search()。有关这些方法的详细信息，请参阅第 121 页的“[在字符串中查找模式并替换子字符串](#)”。

创建正则表达式实例

有两种方法可以创建正则表达式实例。一种方法是使用正斜杠字符 (/) 来界定正则表达式，另一种是使用 new 构造函数。例如，以下两个正则表达式是等效的：

```
var pattern1:RegExp = /bob/i;
var pattern2:RegExp = new RegExp("bob", "i");
```

正斜杠界定正则表达式的方式与用引号界定字符串文本的方式相同。正斜杠内的正则表达式部分定义“模式”。正则表达式还可以在后一个界定斜杠后包含“标志”。这些标志也看作是正则表达式的一部分，但是它们独立于模式。

使用 new 构造函数时，使用两个字符串来定义正则表达式。第一个字符串定义模式，第二个字符串定义标志，如下例所示：

```
var pattern2:RegExp = new RegExp("bob", "i");
```

如果在使用正斜杠界定符定义的正则表达式中包含正斜杠，则必须在正斜杠前面加上反斜杠 (\) 转义字符。例如，下面的正则表达式与模式 1/2 匹配：

```
var pattern:RegExp = /1\//;
```

若要在使用 new 构造函数定义的正则表达式中包含引号，您必须在引号前面加上反斜杠 (\) 转义字符（就像定义任何 String 文本一样）。例如，下面的正则表达式与模式 eat at "joe's" 匹配：

```
var pattern1:RegExp = new RegExp("eat at \"joe's\"", "");
var pattern2:RegExp = new RegExp('eat at "joe\'s"', "");
```

请勿在使用正斜杠界定符定义的正则表达式中对引号使用反斜杠转义字符。同样地，不要在使用 new 构造函数定义的正则表达式中对正斜杠使用转义字符。下列正则表达式是等效的，它们都定义了模式 1/2 "joe's"：

```
var pattern1:RegExp = /1\// "joe's"/;
var pattern2:RegExp = new RegExp('1\// "joe\'s"', "");
var pattern3:RegExp = new RegExp('1\// "joe\'s"', '');
```

此外，在使用 new 构造函数定义的正则表达式中，若要使用以反斜杠 (\) 字符开头的元序列（例如，\d，用于匹配任何数字），请键入两个反斜杠字符：

```
var pattern:RegExp = new RegExp("\\\d+", ""); // matches one or more digits
```

在本实例中您必须键入两个反斜杠字符，因为 RegExp() 构造函数方法的第一个参数是字符串，而在字符串文本中必须键入两个反斜杠字符才能将其识别为单个反斜杠字符。

后面几节将介绍定义正则表达式模式的语法。

有关标志的详细信息，请参阅第 180 页的“[标志和属性](#)”。

字符、元字符和元序列

最简单的正则表达式是与字符序列匹配的表达式，如以下示例中所示：

```
var pattern:RegExp = /hello/;
```

但是，下列字符（称为元字符）在正则表达式中具有特殊含义：

```
^ $ \ . * + ? ( ) [ ] { } |
```

例如，下面的正则表达式所匹配的是字母 A 后跟字母 B 的零个或多个实例（星号元字符表示重复）再跟字母 C：

```
/AB*C/
```

在正则表达式模式中包含元字符时若要使其不具有特殊含义，您必须使用反斜杠 (\) 转义字符。例如，下面的正则表达式与顺序依次为字母 A、字母 B、星号和字母 C 的模式匹配：

```
var pattern:RegExp = /AB\*C/;
```

“元序列”与元字符类似，在正则表达式中具有特殊含义。元序列由多个字符组成。以下几节提供了有关使用元字符和元序列的详细信息。

关于元字符

下表总结了可以在正则表达式中使用的元字符：

元字符	描述
^ (尖号)	匹配字符串的开头。设置 m (multiline) 标志后，尖号还匹配行的开头（请参阅第 180 页的“ 标志和属性 ”）。请注意，尖号用在字符类的开头时表示符号反转而非字符串的开头。有关详细信息，请参阅第 176 页的“ 字符类 ”。
\$ (美元符号)	匹配字符串的结尾。设置 m (multiline) 标志后，\$ 还匹配换行 (\n) 字符前面的位置。有关详细信息，请参阅第 180 页的“ 标志和属性 ”。
\(反斜杠)	对特殊字符的特殊元字符含义进行转义。 此外，如果要在正则表达式文本中使用正斜杠字符，也要使用反斜杠字符，例如，/1\2/ 匹配字符 1 后跟正斜杠字符和字符 2。
.(点)	匹配任意单个字符。 只有设置 s (dotall) 标志时，点才匹配换行字符 (\n)。有关详细信息，请参阅第 180 页的“ 标志和属性 ”。
* (星号)	匹配前面重复零次或多次的项目。 有关详细信息，请参阅第 177 页的“ 数量表示符 ”。
+ (加号)	匹配前面重复一次或多次的项目。 有关详细信息，请参阅第 177 页的“ 数量表示符 ”。
? (问号)	匹配前面重复零次或一次的项目。 有关详细信息，请参阅第 177 页的“ 数量表示符 ”。

元字符	描述
(和)	<p>在正则表达式中定义组。以下情况下使用组：</p> <ul style="list-style-type: none"> 限制逻辑“或”字符 的范围：/(a b c)d/ 定义数量表示符的范围：/(walla.){1,2}/ 用在逆向引用中。例如，下面的正则表达式中的 \1 匹配模式的第一个括号组中的匹配内容： /(\w*) is repeated: \1/ <p>有关详细信息，请参阅第 178 页的“组”。</p>
[和]	<p>定义字符类，字符类定义单个字符可能的匹配：</p> <p>/[aeiou]/ 匹配所指定字符中的任意一个。</p> <p>在字符类中，使用连字符 (-) 指定字符的范围：</p> <p>/[A-Z0-9]/ 匹配从 A 到 Z 的大写字母或 0 到 9 的数字。</p> <p>在字符类中，插入反斜杠对] 和</p> <p>- 字符：</p> <p>/[+\-]\d+/ 匹配一个或多个数字前面的 + 或 -。</p> <p>在字符类中，以下字符（通常为元字符）被看作一般字符（非元字符），不需要反斜杠：</p> <p>/[\$]/£ 匹配 \$ 或 £。</p> <p>有关详细信息，请参阅第 176 页的“字符类”。</p>
(坚线)	<p>用于逻辑“或”操作，匹配左侧或右侧的部分：</p> <p>/abc xyz/ 匹配 abc 或 xyz。</p>

关于元序列

元序列是在正则表达式模式中具有特殊含义的字符序列。下表说明了这些元序列：

元序列	描述
{n}	指定前一项目的数值数量或数量范围：
{n,}	/A{27}/ 匹配重复 27 次的字符 A。
和	/A{3,}/ 匹配重复 3 次或更多次的字符 A。
{n,n}	/A{3,5}/ 匹配重复 3 到 5 次的字符 A。
	有关详细信息，请参阅第 177 页的“ 数量表示符 ”。
\b	匹配单词字符和非单词字符之间的位置。如果字符串中的第一个或最后一个字符是单词字符，则也匹配字符串的开头或结尾。
\B	匹配两个单词字符之间的位置。也匹配两个非单词字符之间的位置。
\d	匹配十进制数字。
\D	匹配除数字以外的任何字符。
\f	匹配换页符。
\n	匹配换行符。
\r	匹配回车符。
\s	匹配任何空白字符（空格、制表符、换行符或回车符）。

元序列	描述
\S	匹配除空白字符以外的任何字符。
\t	匹配制表符。
\unnnn	匹配字符代码由十六进制数字 nnnn 指定的 Unicode 字符。例如，\u263a 是一个笑脸字符。
\v	匹配垂直换页符。
\w	匹配单词字符（A-Z、a-z、0-9 或 _）。请注意，\w 不匹配非英文字符，如 é、ñ 或 ç。
\W	匹配除单词字符以外的任何字符。
\xnn	匹配具有指定 ASCII 值（由十六进制数字 nn 定义）的字符。

字符类

可以使用字符类指定字符列表以匹配正则表达式中的一个位置。使用方括号 ([和]) 定义字符类。例如，下面的正则表达式定义了匹配 bag、beg、big、bog 或 bug 的字符类：

```
/b[aeiou]g/
```

字符类中的转义序列

通常在正则表达式中具有特殊含义的大多数元字符和元序列在字符类中“不具有”那些特殊含义。例如，在正则表达式中星号用于表示重复，但是出现在字符类中时则不具有此含义。下列字符类匹配星号本身以及列出的任何其它字符：

```
/[abc*123]/
```

但是，下表中列出的三个字符功能与元字符相同，在字符类中具有特殊含义：

元字符	在字符类中的含义
]	定义字符类的结尾。
-	定义字符范围（请参阅后面的“字符类中字符的范围”一节）。
\	定义元序列并撤销元字符的特殊含义。

对于要识别为字面字符（无特殊元字符含义）的任何字符，必须在该字符前面加反斜杠转义字符。例如，下面的正则表达式包含匹配四个符号（\$、\、] 或 -）中任意一个符号的字符类。

```
/$\\[-]/
```

除能够保持特殊含义的元字符外，下列元序列在字符类中也具有元序列功能：

元序列	在字符类中的含义
\n	匹配换行符。
\r	匹配回车符。
\t	匹配制表符。
\unnnn	匹配具有指定 Unicode 代码点值（由十六进制数字 nnnn 定义）的字符。
\xnn	匹配具有指定 ASCII 值（由十六进制数字 nn 定义）的字符。

其它正则表达式元序列和元字符在字符类中看作普通字符。

字符类中字符的范围

使用连字符指定字符的范围，例如 A-Z、a-z 或 0-9。这些字符必须在字符类中构成有效的范围。例如，下面的字符类匹配 a-z 范围内的任何一个字符或任何数字：

```
/ [a-zA-Z] /
```

您还可以使用 \xnn ASCII 字符代码通过 ASCII 值指定范围。例如，下面的字符类匹配扩展 ASCII 字符集中的任意字符（如 é 和 ê）：

```
\x
```

反转的字符类

如果在字符类的开头使用尖号 (^) 字符，则将反转该集合的意义，即未列出的任何字符都认为匹配。下面的字符类匹配除小写字母 (a-z) 或数字以外的任何字符：

```
/ [^a-zA-Z] /
```

必须在字符类的开头键入尖号 (^) 字符以表示反转。否则，您只是将尖号字符添加到字符类的字符中。例如，下面的字符类匹配许多符号字符中的任意一个，其中包括尖号：

```
/ [!.,#+*%$&^] /
```

数量表示符

使用数量表示符指定字符或序列在模式中的重复次数，如下所示：

数量表示符元字符	描述
*	匹配前面重复零次或多次的项目。
+	匹配前面重复一次或多次的项目。
?	匹配前面重复零次或一次的项目。
{n}	指定前一项目的数值数量或数量范围：
{n,}	/A{27}/ 匹配重复 27 次的字符 A。
和	/A{3,}/ 匹配重复 3 次或更多次的字符 A。
{n,n}	/A{3,5}/ 匹配重复 3 到 5 次的字符 A。

您可以将数量表示符应用到单个字符、字符类或组：

- /a+/ 匹配重复一次或多次的字符 a。
- /\d+/ 匹配一个或多个数字。
- /[abc]+/ 匹配重复的一个或多个字符，这些字符可能是 a、b 或 c 中的某个。
- /(very,)*/ 匹配重复零次或多次的后跟逗号和空格的单词 very。

您可以在应用数量表示符的括号组内使用数量表示符。例如，下面的数量表示符匹配诸如 word 和 word-word-word 的字符串：

```
/\w+(-\w+)* /
```

默认情况下，正则表达式执行所谓的“无限匹配”。正则表达式中的任何子模式（如 .*）都会尝试在字符串中匹配尽可能多的字符，然后再执行正则表达式的下一部分。例如，使用以下正则表达式和字符串：

```
var pattern:RegExp = /<p>.*</p>/;
str:String = "<p>Paragraph 1</p> <p>Paragraph 2</p>";
```

正则表达式匹配整个字符串：

```
<p>Paragraph 1</p> <p>Paragraph 2</p>
```

但是，假如您只想匹配一个 `<p>...</p>` 组。则可以通过以下操作实现：

```
<p>Paragraph 1</p>
```

在所有数量表示符后添加问号 (?) 以将其更改为所谓的“惰性数量表示符”。例如，下面的正则表达式使用惰性数量表示符 *? 匹配 `<p>` 后跟数量最少（惰性）的字符，再跟 `</p>` 的模式：

```
/<p>.*?</p>/
```

有关数量表示符，请牢记以下几点：

- 数量表示符 {0} 和 {0,0} 不会从匹配中排除项目。
- 不要结合使用多个数量表示符，例如 `/abc+*/` 中。
- 在除非设置 s (dotall) 标志，否则不会跨越多行，即使后跟 * 数量表示符。例如，请考虑使用以下代码：

```
var str:String = "<p>Test\n";
str += "Multiline</p>";
var re:RegExp = /<p>.*</p>/;
trace(str.match(re)); // null;

re = /<p>.*</p>/s;
trace(str.match(re));
// output: <p>Test
//           Multiline</p>
```

有关详细信息，请参阅第 180 页的“[标志和属性](#)”。

逻辑“或”

在正则表达式中使用 | (竖线) 字符可使正则表达式引擎考虑其它匹配。例如，下面的正则表达式匹配单词 cat、dog、pig 和 rat 中的任意一个：

```
var pattern:RegExp = /cat|dog|pig|rat/;
```

您可以使用括号定义组以限制逻辑“或”字符 | 的范围。下面的正则表达式匹配 cat 后跟 nap 或 nip：

```
var pattern:RegExp = /cat(nap|nip)/;
```

有关详细信息，请参阅第 178 页的“[组](#)”。

下面两个正则表达式是等效的，一个使用 | 逻辑“或”字符，另一个使用字符类（由 [和] 定义）：

```
/1|3|5|7|9/
/[13579]/
```

有关详细信息，请参阅第 176 页的“[字符类](#)”。

组

您可以使用括号在正则表达式中指定组，如下所示：

```
/class-(\d*)/
```

组是模式的子部分。您可以使用组实现以下操作：

- 将数量表示符应用到多个字符。
- 界定要应用逻辑“或”（通过使用 | 字符）的子模式。
- 捕获正则表达式中的子字符串匹配（例如，在正则表达式中使用 \1 以匹配先前匹配的组，或类似地在 String 类的 replace() 方法中使用 \$1）。

下面几节将介绍有关这些组用法的详细信息。

使用带数量表示符的组

如果不使用组，数量表示符将应用到它前面的字符或字符类，如下所示：

```
var pattern:RegExp = /ab*/ ;
// matches the character a followed by
// zero or more occurrences of the character b

pattern = /a\d+/";
// matches the character a followed by
// one or more digits

pattern = /a[123]{1,3}/;
// matches the character a followed by
// one to three occurrences of either 1, 2, or 3
```

然而，您可以使用组将数量表示符应用到多个字符或字符类：

```
var pattern:RegExp = /(ab)*/;
// matches zero or more occurrences of the character a
// followed by the character b, such as ababab

pattern = /(a\d)+/;
// matches one or more occurrences of the character a followed by
// a digit, such as a1a5a8a3

pattern = /(spam ){1,3}/;
// matches 1 to 3 occurrences of the word spam followed by a space
```

有关数量表示符的详细信息，请参阅第 177 页的“[数量表示符](#)”。

使用带逻辑“或”字符 (|) 的组

可以使用组来定义一组要应用逻辑“或”字符 (|) 的字符，如下所示：

```
var pattern:RegExp = /cat|dog/;
// matches cat or dog

pattern = /ca(t|d)og/;
// matches catog or cadog
```

使用组捕获子字符串匹配

如果您在模式中定义标准括号组，则之后可以在正则表达式中引用它。这称为“逆向引用”，并且此类型的组称为“捕获组”。

例如，在下面的正则表达式中，序列 \1 匹配在捕获括号组中匹配的任意子字符串：

```
var pattern:RegExp = /(\d+)-by-\1/;
// matches the following: 48-by-48
```

您可以通过键入 \1、\2、...、\99 在正则表达式中最多指定 99 个此类逆向引用。

类似地，在 String 类的 replace() 方法中，可以使用 \$1-\$99 在替换字符串中插入捕获的组子字符串匹配：

```
var pattern:RegExp = /Hi, (\w+)\./;
var str:String = "Hi, Bob.";
trace(str.replace(pattern, "$1, hello."));
// output: Bob, hello.
```

此外，如果使用捕获组， RegExp 类的 exec() 方法和 String 类的 match() 方法将返回与捕获组匹配的子字符串：

```
var pattern:RegExp = /(\w+)@(\w+)\.(\w+)/;
var str:String = "bob@example.com";
trace(pattern.exec(str));
// bob@example.com,bob@example.com
```

使用非捕获组和向前查找组

非捕获组是只用于分组的组，它不会被“收集”，也不会匹配有限的逆向引用。可以使用 (?: 和) 来定义非捕获组，如下所示：

```
var pattern = /(?:com|org|net);
```

例如，注意在捕获组和非捕获组中加入 (com|org) 的区别 (exec() 方法在完全匹配后列出捕获组)：

```
var pattern:RegExp = /(\w+)@(\w+).(com|org)/;
var str:String = "bob@example.com";
trace(pattern.exec(str));
// bob@example.com,bob@example,com
```

//noncapturing:

```
var pattern:RegExp = /(\w+)@(\w+).(?:com|org)/;
var str:String = "bob@example.com";
trace(pattern.exec(str));
// bob@example.com,bob@example
```

一类特殊的非捕获组是“向前查找组”，它包括两种类型：“正向前查找组”和“负向前查找组”。

使用 (?= 和) 定义正向前查找组，它指定组中的子模式位置必须匹配。但是，匹配正向前查找组的字符串部分可能匹配正则表达式中的剩余模式。例如，由于 (?=e) 在下面的代码中是正向前查找组，它匹配的字符 e 可以被正则表达式的后续部分匹配，在本例中为捕获组 \w*：

```
var pattern:RegExp = /sh(?=e)(\w*)/i;
var str:String = "Shelly sells seashells by the seashore";
trace(pattern.exec(str));
// Shelly,elly
```

使用 (?! 和) 定义负向前查找组，它指定该组中的子模式位置不得匹配。例如：

```
var pattern:RegExp = /sh(?!=e)(\w*)/i;
var str:String = "She sells seashells by the seashore";
trace(pattern.exec(str));
// shore,ore
```

使用命名组

命名组是正则表达式中给定命名标识符的一类组。使用 (?P<name> 和) 可定义命名组。例如，下面的正则表达式包含标识符命名为 digits 的命名组：

```
var pattern = /[a-z]+(?P<digits>\d+)[a-z]+/;
```

如果使用 exec() 方法，将添加一个匹配的命名组作为 result 数组属性：

```
var myPattern:RegExp = /([a-z]+)(?P<digits>\d+)[a-z]+/;
var str:String = "a123bcd";
var result:Array = myPattern.exec(str);
trace(result.digits); // 123
```

这里还有一个例子，它使用两个命名组，标识符分别为 name 和 dom：

```
var emailPattern:RegExp =
    /(?P<name>(\w|[_\.-])+)(?P<dom>((\w|-)+))+.\w{2,4}+/";
var address:String = "bob@example.com";
var result:Array = emailPattern.exec(address);
trace(result.name); // bob
trace(result.dom); // example
```

注：命名组不属于 ECMAScript 语言规范。它们是 ActionScript 3.0 中的新增功能。

标志和属性

下表列出了可以为正则表达式设置的五种标志。每种标志都可以作为正则表达式对象属性进行访问。

标志	属性	描述
g	global	匹配多个匹配。
i	ignoreCase	不区分大小写的匹配。应用于 A-Z 和 a-z 字符，但不能应用于扩展字符，如 É 和 é。
m	multiline	设置此标志后，\$ 和 ^ 可以分别匹配行的开头和结尾。
s	dotall	设置此标志后，.（点）可以匹配换行符 (\n)。
x	extended	允许扩展的正则表达式。您可以在正则表达式中键入空格，它将作为模式的一部分被忽略。这可使您更加清晰可读地键入正则表达式代码。

请注意这些属性都是只读属性。您在设置正则表达式变量时，可以设置标志（g、i、m、s 和 x），如下所示：

```
var re:RegExp = /abc/gimsx;
```

但是，您无法直接设置命名属性。例如，下列代码将导致错误：

```
var re:RegExp = /abc/;
re.global = true; // This generates an error.
```

默认情况下，除非您在正则表达式声明中指定这些标志，否则不会设置，并且相应的属性也会设置为 false。

另外，还有其它两种正则表达式属性：

- lastIndex 属性指定字符串中的索引位置以用于下次调用正则表达式的 exec() 或 test() 方法。
- source 属性指定定义正则表达式的模式部分的字符串。

g (global) 标志

如果不包含 g (global) 标志，正则表达式匹配将不超过一个。例如，如果正则表达式中不包含 g 标志，String.match() 方法只返回一个匹配的子字符串：

```
var str:String = "she sells seashells by the seashore.";
var pattern:RegExp = /sh\w*/;
trace(str.match(pattern)) // output: she
```

如果设置 g 标志，String.match() 方法将返回多个匹配，如下所示：

```
var str:String = "she sells seashells by the seashore.";
var pattern:RegExp = /sh\w*/g;
// The same pattern, but this time the g flag IS set.
trace(str.match(pattern)); // output: she,shells,shore
```

i (ignoreCase) 标志

默认情况下，正则表达式匹配区分大小写。如果设置 i (ignoreCase) 标志，将忽略区分大小写。例如，正则表达式中的小写字母 s 不会匹配大写字母 S（字符串中的第一个字符）：

```
var str:String = "She sells seashells by the seashore.";
trace(str.search(/sh/)); // output: 13 -- Not the first character
```

但是如果设置 i 标志，正则表达式将匹配大写字母 S：

```
var str:String = "She sells seashells by the seashore.";
trace(str.search(/sh/i)); // output: 0
```

i 标志仅忽略 A-Z 和 a-z 字符的大小写，而不忽略扩展字符的大小写，如 É 和 é。

m (multiline) 标志

如果没有设置 m (multiline) 标志，^ 将匹配字符串的开头，而 \$ 匹配字符串的结尾。如果设置 m 标志，这些字符将分别匹配行的开头和结尾。请考虑使用下列包含换行符的字符串：

```
var str:String = "Test\n";
str += "Multiline";
trace(str.match(/^\w*/g)); // Match a word at the beginning of the string.
```

即使在正则表达式中设置 g (global) 标志，match() 方法也会只匹配一个子字符串，因为对于 ^ (字符串开头) 只有一个匹配。输出结果为：

```
Test
```

下面是设置了 m 标志的同一段代码：

```
var str:String = "Test\n";
str += "Multiline";
trace(str.match(/^\w*/gm)); // Match a word at the beginning of lines.
```

这次输出结果同时包含两行开头的单词：

```
Test,Multiline
```

请注意，只有 \n 字符表示行的结束。下列字符不表示行的结束：

- 回车 (\r) 字符
- Unicode 行分隔符 (\u2028) 字符
- Unicode 段分隔符 (\u2029) 字符

s (dotall) 标志

如果没有设置 s (dotall 或 “dot all”) 标志，则正则表达式中的点(.) 将不匹配换行符 (\n)。因此，下面的示例没有匹配：

```
var str:String = "<p>Test\n";
str += "Multiline</p>";
var re:RegExp = /<p>.*?</p>/;
trace(str.match(re));
```

但是，如果设置了 s 标志，点就匹配换行符：

```
var str:String = "<p>Test\n";
str += "Multiline</p>";
var re:RegExp = /<p>.*?</p>/s;
trace(str.match(re));
```

在本例中，匹配内容是 <p> 标签内的整个子字符串，其中包括换行符：

```
<p>Test
Multiline</p>
```

x (extended) 标志

正则表达式有时很难阅读，特别是当其包含很多元字符和元序列时。例如：

```
/<p>(>|(\s*[^>]*)) .*?</p>/gi
```

在正则表达式中使用 x (extended) 标志时，则会忽略在模式中键入的所有空格。例如，下面的正则表达式同前面的示例相同：

```
/      <p      (> |  (\s*  [^>]* >))    .*?    </p>  /gix
```

如果设置了 x 标志，而且希望匹配空格字符，则应在空格前加上反斜杠。例如，以下两个正则表达式是等效的：

```
/foo bar/
/foo \ bar/x
```

lastIndex 属性

lastIndex 属性在字符串中指定开始进行下一次搜索的索引位置。对于将 g 标志设置为 true 的正则表达式，此属性会影响对该表达式调用的 exec() 和 test() 方法。例如，请考虑使用以下代码：

```

var pattern:RegExp = /p\w*/gi;
var str:String = "Pedro Piper picked a peck of pickled peppers.";
trace(pattern.lastIndex);
var result:Object = pattern.exec(str);
while (result != null)
{
    trace(pattern.lastIndex);
    result = pattern.exec(str);
}

```

默认情况下，`lastIndex` 属性设置为 0（从字符串的开头开始搜索）。每次匹配完成后，都会设为该匹配后的索引位置。因此，前面代码的输出如下所示：

```

0
5
11
18
18
25
36
44

```

如果将 `global` 标志设置为 `false`，则 `exec()` 和 `test()` 方法不会使用或设置 `lastIndex` 属性。

`String` 类的 `match()`、`replace()` 和 `search()` 方法都从字符串的开头进行搜索，而不考虑调用该方法时所使用的正则表达式中的 `lastIndex` 属性设置。（但是，`match()` 方法不会将 `lastIndex` 设为 0。）

可以设置 `lastIndex` 属性来调整字符串中的起始位置以对正则表达式进行匹配。

source 属性

`source` 属性指定用于定义正则表达式的模式部分的字符串。例如：

```

var pattern:RegExp = /foo/gi;
trace(pattern.source); // foo

```

对字符串使用正则表达式的方法

`RegExp` 类包含两个方法：`exec()` 和 `test()`。

除 `RegExp` 类的 `exec()` 和 `test()` 方法外，`String` 类还包含以下方法，使您可以在字符串中匹配正则表达式：`match()`、`replace()`、`search()` 和 `splice()`。

test() 方法

`RegExp` 类的 `test()` 方法只检查提供的字符串是否包含正则表达式的匹配内容，如下面的示例所示：

```

var pattern:RegExp = /Class-\w/;
var str = "Class-A";
trace(pattern.test(str)); // output: true

```

exec() 方法

`RegExp` 类的 `exec()` 方法检查提供的字符串是否有正则表达式的匹配，并返回具有如下内容的数组：

- 匹配的子字符串
- 同正则表达式中的任意括号组匹配的子字符串

该数组还包含 `index` 属性，此属性指明子字符串匹配起始的索引位置。

例如，请考虑使用以下代码：

```
var pattern:RegExp = /\d{3}[-]\d{3}-\d{4}/; //U.S phone number
var str:String = "phone: 415-555-1212";
var result:Array = pattern.exec(str);
trace(result.index, " - ", result);
// 7-415-555-1212
```

在正则表达式设置了 g (global) 标志时，多次使用 exec() 方法可以匹配多个子字符串：

```
var pattern:RegExp = /\w*sh\w*/gi;
var str:String = "She sells seashells by the seashore";
var result:Array = pattern.exec(str);

while (result != null)
{
    trace(result.index, "\t", pattern.lastIndex, "\t", result);
    result = pattern.exec(str);
}
//output:
// 0 3 She
// 10 19 seashells
// 27 35 seashore
```

使用 RegExp 参数的 String 方法

下列 String 类方法将正则表达式作为参数：match()、replace()、search() 和 split()。有关这些方法的详细信息，请参阅第 121 页的“[在字符串中查找模式并替换子字符串](#)”。

示例：Wiki 分析器

这个简单的 Wiki 文本转换示例说明了正则表达式的一些用途：

- 将与源 Wiki 模式匹配的文本行转换为相应的 HTML 输出字符串。
- 使用正则表达式将 URL 模式转换为 HTML <a> 超链接标签。
- 使用正则表达式将美元符号字符串（如 "\$9.95"）转换为欧元符号字符串（如 "8.24 €"）。

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。WikiEditor 应用程序文件位于文件夹 Samples/WikiEditor 中。该应用程序包含以下文件：

文件	描述
WikiEditor.mxml 或 WikiEditor.fla	Flash 或 Flex 中的主应用程序文件（分别为 FLA 和 MXML）。
com/example/programmingas3/regExpExamples/WikiParser.as	该类包含使用正则表达式将 Wiki 输入文本模式转换为等效 HTML 输出的方法。
com/example/programmingas3/regExpExamples/UrlParser.as	该类包含使用正则表达式将 URL 字符串转换为 HTML <a> 超链接标签的方法。
com/example/programmingas3/regExpExamples/CurrencyConverter.as	该类包含使用正则表达式将美元符号字符串转换为欧元符号字符串的方法。

定义 WikiParser 类

WikiParser 类包含将 Wiki 输入文本转换为等效 HTML 输出的方法。它虽然不是功能非常强大的 Wiki 转换应用程序，但是说明了正则表达式在模式匹配和字符串转换方面的一些很好的用法。

构造函数和 setWikiData() 方法一起，可简单地初始化 Wiki 输入文本范例字符串，如下所示：

```
public function WikiParser()
{
    wikiData = setWikiData();
}
```

当用户单击范例应用程序中的“Test”按钮时，应用程序将调用 WikiParser 对象的 parseWikiString() 方法。此方法可调用许多其它方法，这些方法再组合输出的 HTML 字符串。

```
public function parseWikiString(wikiString:String):String
{
    var result:String = parseBold(wikiString);
    result = parseItalic(result);
    result = linesToParagraphs(result);
    result = parseBullets(result);
    return result;
}
```

所调用的每个方法（parseBold()、parseItalic()、linesToParagraphs() 和 parseBullets()）都会使用字符串的 replace() 方法替换由正则表达式定义的匹配模式，以便将 Wiki 输入文本转换为 HTML 格式的文本。

转换粗体和斜体模式。

parseBold() 方法会查找 Wiki 粗体文本模式（如 "foo"）并将其转换为等效的 HTML 格式（如 foo），如下所示：

```
private function parseBold(input:String):String
{
    var pattern:RegExp = /'*(.*?)*'/g;
    return input.replace(pattern, "<b>$1</b>");
}
```

请注意，正则表达式的 (.?) 部分匹配两个定义 "" 模式之间任意数目的字符 (*)。下面的 ? 数量表示符使匹配不再是无限制的，因此对于字符串 "aaa" bbb "ccc"，第一个匹配的字符串是 "aaa" 而不是以 "" 模式开头和结尾的整个字符串。

正则表达式中的括号定义捕获组，replace() 方法通过在替换字符串中使用 \$1 代码引用此组。正则表达式中的 g (global) 标志确保 replace() 方法替换字符串中的所有匹配（不仅仅是第一个）。

parseItalic() 方法的原理与 parseBold() 方法类似，只是前者检查作为斜体文本分隔符的两个省略号 ("")（而不是三个）：

```
private function parseItalic(input:String):String
{
    var pattern:RegExp = /'*(.*?)*'/g;
    return input.replace(pattern, "<i>$1</i>");
}
```

转换项目符号模式

如下面的示例所示，parseBullet() 方法会查找 Wiki 项目符号行模式（如 * foo）并将其转换为等效的 HTML 格式（如 foo）：

```
private function parseBullets(input:String):String
{
    var pattern:RegExp = /^\\*(.*?)/gm;
    return input.replace(pattern, "<li>$1</li>");
}
```

正则表达式开头的 ^ 符号匹配行的开头。正则表达式中的 m (multiline) 标志使正则表达式用 ^ 符号来匹配行的开头，而不是简单地匹配字符串的开头。

* 模式匹配星号字符（反斜杠用于表示星号本身，而不是 * 数量表示符）。

正则表达式中的括号定义捕获组，`replace()` 方法通过在替换字符串中使用 `$1` 代码引用此组。正则表达式中的 `g (global)` 标志确保 `replace()` 方法替换字符串中的所有匹配（不仅仅是第一个）。

转换段落 Wiki 模式

`linesToParagraphs()` 方法将每行中输入的 Wiki 字符串转换为 HTML `<p>` 段落标签。该方法中的这些行从输入的 Wiki 字符串中去除空行：

```
var pattern:RegExp = /^$/gm;
var result:String = input.replace(pattern, "");
```

正则表达式中的 `^` 和 `$` 符号分别匹配行的开头和结尾。正则表达式中的 `m (multiline)` 标志使正则表达式用 `^` 符号来匹配行的开头，而不是简单地匹配字符串的开头。

`replace()` 方法使用空字符串（""）替换所有匹配子字符串（空行）。正则表达式中的 `g (global)` 标志确保 `replace()` 方法替换字符串中的所有匹配（不仅仅是第一个）。

将 URL 转换为 HTML `<a>` 标记

当用户单击范例应用程序中的“Test”按钮时，如果用户选中了 `urlToATag` 复选框，应用程序将调用 `URLParser.urlToATag()` 静态方法以将 URL 字符串从输入的 Wiki 字符串转换为 HTML `<a>` 标签。

```
var protocol:String = "((?:http|ftp)://)";
var urlPart:String = "([a-z0-9_-]+\. [a-z0-9_-]+)";
var optionalUrlPart:String = "(\. [a-z0-9_-]* )";
var urlPattern:RegExp = new RegExp(protocol + urlPart + optionalUrlPart, "ig");
var result:String = input.replace(urlPattern, "<a href='$1$2$3'><u>$1$2$3</u></a>");
```

`RegExp()` 构造函数用于将各组成部分组合成正则表达式（`urlPattern`）。这些组成部分是定义正则表达式模式部分的各个字符串。

由 `protocol` 字符串定义的正则表达式模式的第一部分定义了 URL 协议：`http://` 或 `ftp://`。括号定义了由 `?` 符号表示的非捕获组。这意味着括号只是用来定义用于 `|` 逻辑“或”模式的组，该组不会匹配 `replace()` 方法的替换字符串中的逆向引用代码（`$1、$2、$3`）。

正则表达式的其它组成部分都会使用捕获组（由模式中的括号表示），然后用在 `replace()` 方法的替换字符串中的逆向引用代码（`$1、$2、$3`）中。

由 `urlPart` 字符串定义的模式部分至少匹配下列字符中的一个：`a-z、0-9、_` 或 `-`。`+` 数量表示符表明至少一个字符匹配。下面的 `\.` 表示一个必需的点（`.`）字符。其余部分匹配至少包含以下字符中的一个的字符串：`a-z、0-9、_` 或 `-`。

由 `optionalUrlPart` 字符串定义的模式部分匹配零个或多个以下字符：点（`.`）字符后跟任意数目的字母数字字符（包括 `_` 和 `-`）。`*` 数量表示符表明零个或多个字符匹配。

调用 `replace()` 方法可应用正则表达式，并且使用逆向引用组合替换 HTML 字符串。

然后 `urlToATag()` 方法会调用 `emailToATag()` 方法，后者使用类似的技术用 HTML `<a>` 超链接字符串替换电子邮件模式。在本范例文件中用于匹配 HTTP、FTP 和电子邮件 URL 的正则表达式是相当简单的，只是起到示范作用，还有更复杂的正则表达式更准确地匹配这类 URL。

将美元符号字符串转换为欧元符号字符串

当用户单击范例应用程序中的“Test”按钮时，如果用户选中了 `dollarToEuro` 复选框，应用程序将调用 `CurrencyConverter.usdToEuro()` 静态方法以将美元符号字符串（如 `"$9.95"`）转换为欧元符号字符串（如 `"8.24 €"`），如下所示：

```
var usdPrice:RegExp = /\$([\d,]+\.\d+)/g;
return input.replace(usdPrice, usdStrToEuroStr);
```

第一行定义的简单模式匹配美元符号字符串。请注意，`$` 字符前面加反斜杠（`\`）转义字符。

replace() 方法将正则表达式用作模式匹配参数，并调用 usdStrToEuroStr() 函数以确定替换字符串（欧元值）。

如果将函数名称用作 replace() 方法的第二个参数时，则会将以下内容作为参数传递给被调用的函数：

- 字符串的匹配部分。
- 任何捕获的括号组匹配。按这种方式传递的参数数目因捕获的括号组匹配的数目而异。您可以通过检查函数代码中的 arguments.length - 3 来确定捕获的括号组匹配的数目。
- 字符串中匹配开始的索引位置。
- 完整的字符串。

usdStrToEuroStr() 方法将美元符号字符串模式转换为欧元符号字符串，如下所示：

```
private function usdToEuro(...args):String
{
    var usd:String = args[1];
    usd = usd.replace(",","");
    var exchangeRate:Number = 0.828017;
    var euro:Number = Number(usd) * exchangeRate;
    trace(usd, Number(usd), euro);
    const euroSymbol:String = String.fromCharCode(8364); // €
    return euro.toFixed(2) + " " + euroSymbol;
}
```

请注意，args[1] 表示由 usdPrice 正则表达式匹配的捕获括号组。这是美元符号字符串的数字部分，也就是没有 \$ 符号的美元数目。该方法应用汇率转换并返回生成的字符串（带有尾随符号 €，而不是前导符号 \$）。

第 11 章：处理 XML

ActionScript 3.0 包含一组基于 ECMAScript for XML (E4X) 规范 (ECMA-357 第 2 版) 的类。这些类包含用于处理 XML 数据的强大且易用的功能。与以前的编程技术相比，使用 E4X 可以更快地用 XML 数据开发代码。此外，开发的代码也更容易阅读。

本章介绍如何使用 E4X 来处理 XML 数据。

XML 基础知识

处理 XML 简介

XML 是一种表示结构化信息的标准方法，以使计算机能够方便地使用此类信息，并且人们可以非常方便地编写和理解这些信息。XML 是 eXtensible Markup Language (可扩展标记语言) 的缩写。www.w3.org/XML/ 上提供了 XML 标准。

XML 提供了一种简便的标准方法对数据进行分类，以使其更易于读取、访问以及处理。XML 使用类似于 HTML 的树结构和标签结构。以下是一个简单的 XML 数据示例：

```
<song>
    <title>What you know?</title>
    <artist>Steve and the flubberblubs</artist>
    <year>1989</year>
    <lastplayed>2006-10-17-08:31</lastplayed>
</song>
```

XML 数据也可能会比较复杂，其中包含嵌套在其它标签中的标签以及属性和其它结构组件。以下是一个比较复杂的 XML 数据示例：

```
<album>
    <title>Questions, unanswered</title>
    <artist>Steve and the flubberblubs</artist>
    <year>1989</year>
    <tracks>
        <song tracknumber="1" length="4:05">
            <title>What do you know?</title>
            <artist>Steve and the flubberblubs</artist>
            <lastplayed>2006-10-17-08:31</lastplayed>
        </song>
        <song tracknumber="2" length="3:45">
            <title>Who do you know?</title>
            <artist>Steve and the flubberblubs</artist>
            <lastplayed>2006-10-17-08:35</lastplayed>
        </song>
        <song tracknumber="3" length="5:14">
            <title>When do you know?</title>
            <artist>Steve and the flubberblubs</artist>
            <lastplayed>2006-10-17-08:39</lastplayed>
        </song>
        <song tracknumber="4" length="4:19">
            <title>Do you know?</title>
            <artist>Steve and the flubberblubs</artist>
            <lastplayed>2006-10-17-08:44</lastplayed>
        </song>
    </tracks>
</album>
```

请注意，此 XML 文档中包含其它完整 XML 结构（如 `song` 标签及其子标签）。它还说明了其它 XML 结构，如属性（`song` 标签中的 `tracknumber` 和 `length`）以及包含其它标签而不是包含数据的标签（如 `tracks` 标签）。

XML 快速入门

如果您没有或几乎没有 XML 方面的经验，您可以阅读下面对 XML 数据的最常见特性的简要说明。XML 数据是以纯文本格式编写的，并使用特定语法将信息组织为结构化格式。通常，将一组 XML 数据称为“XML 文档”。在 XML 格式中，可通过分层结构将数据组织到“元素”（可以是单个数据项，也可以是其它元素的容器）中。每个 XML 文档将一个元素作为顶级项目或主项目；此根元素内可能会包含一条信息，但更可能会包含其它元素，而这些元素又包含其它元素，依此类推。例如，以下 XML 文档包含有关音乐唱片的信息：

```
<song tracknumber="1" length="4:05">
    <title>What do you know?</title>
    <artist>Steve and the flubberblubs</artist>
    <mood>Happy</mood>
    <lastplayed>2006-10-17-08:31</lastplayed>
</song>
```

每个元素都是用一组标签来区分的，即包含在尖括号（小于号和大于号）中的元素名称。开始标签（指示元素的开头）包含元素名称：

```
<title>
```

结束标签（标记元素的结尾）在元素名称前面包含一个正斜杠：

```
</title>
```

如果元素不包含任何内容，则会将其编写为一个空元素（有时称为自结束元素）。在 XML 中，以下元素：

```
<lastplayed/>
```

与下面的元素完全相同：

```
<lastplayed></lastplayed>
```

除了在开始和结束标签之间包含的元素内容外，元素还可以包含在元素开始标签中定义的其它值（称为“属性”）。例如，以下 XML 元素定义一个名为 `length` 且值为“4:19”的属性：

```
<song length="4:19"></song>
```

每个 XML 元素都包含内容，这可以是单个值、一个或多个 XML 元素或没有任何内容（对于空元素）。

了解有关 XML 的详细信息

要了解有关处理 XML 的详细信息，请参阅额外的一些书籍和资源以了解有关 XML 的详细信息，其中包括以下 Web 站点：

- W3Schools XML 教程：<http://w3schools.com/xml/>
- XML.com：<http://www.xml.com/>
- XMLpitstop 教程、讨论列表等等：<http://xmlpitstop.com/>

用于处理 XML 的 ActionScript 类

ActionScript 3.0 包含一些用于处理 XML 结构化信息的类。下面列出了两个主类：

- XML：表示单个 XML 元素，它可以是包含多个子元素的 XML 文档，也可以是文档中的单值元素。
- XMLList：表示一组 XML 元素。当具有多个“同级”（在 XML 文档层次结构中的相同级别，并且包含在相同父级中）的 XML 元素时，将使用 XMLList 对象。例如，XMLList 实例是使用以下一组 XML 元素（可能包含在 XML 文档中）的最简便方法：

```
<artist type="composer">Fred Wilson</artist>
<artist type="conductor">James Schmidt</artist>
<artist type="soloist">Susan Harriet Thurndon</artist>
```

对于涉及 XML 命名空间的更高级用法，ActionScript 还包含 `Namespace` 和 `QName` 类。有关详细信息，请参阅第 201 页的“[使用 XML 命名空间](#)”。

除了用于处理 XML 的内置类外，ActionScript 3.0 还包含一些运算符，它们提供了用于访问和处理 XML 数据的特定功能。这种使用这些类和运算符来处理 XML 的方法称为 `ECMAScript for XML (E4X)`，它是由 ECMA-357 第 2 版规范定义的。

常见 XML 任务

在 ActionScript 中处理 XML 时，您可能会执行以下任务：

- 构造 XML 文档（添加元素和值）
- 访问 XML 元素、值和属性
- 过滤（从中搜索）XML 元素
- 循环访问一组 XML 元素
- 在 XML 类和 `String` 类之间转换数据
- 处理 XML 命名空间
- 加载外部 XML 文件

重要概念和术语

以下参考列表包含将会在本章中使用的重要术语：

- **元素**：XML 文档中的单个项目，它被标识为开始标签和结束标签之间包含的内容（包括标签）。XML 元素可以包含文本数据或其它元素，也可以为空。
- **空元素**：不包含任何子元素的 XML 元素。通常，将空元素编写为自结束标签（如 `<element/>`）。
- **文档**：单个 XML 结构。XML 文档可以包含任意数量的元素（或者仅包含单个空元素）；但是，XML 文档必须具有一个顶级元素，该元素包含文档中的所有其它元素。
- **节点**：XML 元素的另一种称谓。
- **属性**：与元素关联的命名值，它以 `attributename="value"` 格式写入到元素的开始标签中，而不是编写为嵌套在元素内的单独子元素。

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。实质上本章中的代码清单已经包括适当的 `trace()` 函数调用。要测试本章中的代码清单，请执行以下操作：

- 1 创建一个空的 Flash 文档。
- 2 在时间轴上选择一个关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 通过“控制”>“测试影片”运行程序。

可在“输出”面板中看到 `trace()` 函数的结果。

用于测试示例代码清单的此项技术以及其它技术在第 29 页的“[测试本章内的示例代码清单](#)”中有更详细的介绍。

用于处理 XML 的 E4X 方法

ECMAScript for XML 规范定义了一组用于处理 XML 数据的类和功能。这些类和功能统称为 E4X。ActionScript 3.0 包含以下 E4X 类：XML、XMLList、QName 和 Namespace。

E4X 类的方法、属性和运算符旨在实现以下目标：

- 简单 — 在可能的情况下，使用 E4X 可以更容易地编写和理解用于处理 XML 数据的代码。
- 一致 — E4X 背后的方法和推理在内部是一致的，并与 ActionScript 的其它部分保持一致。
- 熟悉 — 使用众所周知的运算符来处理 XML 数据，如点(.) 运算符。

注：ActionScript 2.0 中有一个 XML 类。在 ActionScript 3.0 中，该类已重新命名为 XMLDocument，从而避免与作为 E4X 的一部分的 ActionScript 3.0 XML 类冲突。在 ActionScript 3.0 中，flash.xml 包中包含了 XMLDocument、XMLNode、XMLParser 和 XMLTag 几个旧类，主要是用于旧支持。新的 E4X 类是核心类；无需导入包即可使用这些类。本章不详细阐述旧的 ActionScript 2.0 XML 类。有关这些类的详细信息，请参阅《ActionScript 3.0 语言和组件参考》中的 [flash.xml 包](#)。

下面是使用 E4X 处理数据的一个示例：

```
var myXML:XML =
<order>
  <item id='1'>
    <menuName>burger</menuName>
    <price>3.95</price>
  </item>
  <item id='2'>
    <menuName>fries</menuName>
    <price>1.45</price>
  </item>
</order>
```

通常，应用程序都会从外部源（如 Web 服务或 RSS 供给）加载 XML 数据。不过为清楚起见，本章中的示例是将 XML 数据作为文本来分配的。

如下面的代码所示，E4X 包含了一些直观运算符（如点(.) 和属性标识符(@) 运算符），用于访问 XML 中的属性 (property) 和属性 (attribute)：

```
trace(myXML.item[0].menuName); // Output: burger
trace(myXML.item.(@id==2).menuName); // Output: fries
trace(myXML.item.(menuName=="burger").price); // Output: 3.95
```

使用 appendChild() 方法可为 XML 分配一个新的子节点，如下面的代码片断所示：

```
var newItem:XML =
<item id="3">
  <menuName>medium cola</menuName>
  <price>1.25</price>
</item>

myXML.appendChild(newItem);
```

使用 @ 和 . 运算符不仅可以读取数据，还可以分配数据，如下所示：

```
myXML.item[0].menuName="regular burger";
myXML.item[1].menuName="small fries";
myXML.item[2].menuName="medium cola";

myXML.item.(menuName=="regular burger").@quantity = "2";
myXML.item.(menuName=="small fries").@quantity = "2";
myXML.item.(menuName=="medium cola").@quantity = "2";
```

使用 for 循环可以循环访问 XML 的节点，如下所示：

```

var total:Number = 0;
for each (var property:XML in myXML.item)
{
    var q:int = Number(property.@quantity);
    var p:Number = Number(property.price);
    var itemTotal:Number = q * p;
    total += itemTotal;
    trace(q + " " + property.menuName + " $" + itemTotal.toFixed(2))
}
trace("Total: $" , total.toFixed(2));

```

XML 对象

XML 对象可能表示 XML 元素、属性、注释、处理指令或文本元素。

XML 对象分为包含“简单内容”和包含“复杂内容”两类。有子节点的 XML 对象归入包含复杂内容的一类。如果 XML 对象是属性、注释、处理指令或文本节点之中的任何一个，我们就说它包含简单内容。

例如，下面的 XML 对象包含复杂内容，包括一条注释和一条处理指令：

```

XML.ignoreComments = false;
XML.ignoreProcessingInstructions = false;
var x1:XML =
<order>
    <!--This is a comment. -->
    <?PROC_INSTR sample ?>
    <item id='1'>
        <menuName>burger</menuName>
        <price>3.95</price>
    </item>
    <item id='2'>
        <menuName>fries</menuName>
        <price>1.45</price>
    </item>
</order>

```

如下面的示例所示，现在可以使用 `comments()` 和 `processingInstructions()` 方法创建新的 XML 对象，即注释和处理指令：

```

var x2:XML = x1.comments() [0];
var x3:XML = x1.processingInstructions() [0];

```

XML 属性

XML 类有五个静态属性：

- `ignoreComments` 和 `ignoreProcessingInstructions` 属性确定分析 XML 对象时是否忽略注释或处理指令。
- `ignoreWhitespace` 属性确定在只由空白字符分隔的元素标签和内嵌表达式中是否忽略空白字符。
- `prettyIndent` 和 `prettyPrinting` 属性用于设置由 XML 类的 `toString()` 和 `toXMLString()` 方法返回的文本的格式。

有关这些属性的详细信息，请参阅 [ActionScript 3.0 语言和组件参考](#)。

XML 方法

以下方法用于处理 XML 对象的分层结构：

- `appendChild()`
- `child()`

- childIndex()
- children()
- descendants()
- elements()
- insertChildAfter()
- insertChildBefore()
- parent()
- prependChild()

以下方法用于处理 XML 对象属性 (attribute):

- attribute()
- attributes()

以下方法用于处理 XML 对象属性 (property):

- hasOwnProperty()
- propertyIsEnumerable()
- replace()
- setChildren()

以下方法用于处理限定名和命名空间:

- addNamespace()
- inScopeNamespaces()
- localName()
- name()
- namespace()
- namespaceDeclarations()
- removeNamespace()
- setLocalName()
- setName()
- setNamespace()

以下方法用于处理和确定某些类型的 XML 内容:

- comments()
- hasComplexContent()
- hasSimpleContent()
- nodeKind()
- processingInstructions()
- text()

以下方法用于转换为字符串和设置 XML 对象的格式:

- defaultSettings()
- setSettings()

- settings()
- normalize()
- toString()
- toXMLString()

还有其它几个方法：

- contains()
- copy()
- valueOf()
- length()

有关这些方法的详细信息，请参阅 [ActionScript 3.0 语言和组件参考](#)。

XMLElement 对象

XMLElement 实例表示 XML 对象的任意集合。它可以包含完整的 XML 文档、XML 片断或 XML 查询结果。

以下方法用于处理 XMLElement 对象的分层结构：

- child()
- children()
- descendants()
- elements()
- parent()

以下方法用于处理 XMLElement 对象属性 (attribute)：

- attribute()
- attributes()

以下方法用于处理 XMLElement 属性 (property)：

- hasOwnProperty()
- propertyIsEnumerable()

以下方法用于处理和确定某些类型的 XML 内容：

- comments()
- hasComplexContent()
- hasSimpleContent()
- processingInstructions()
- text()

以下方法用于转换为字符串和设置 XMLElement 对象的格式：

- normalize()
- toString()
- toXMLString()

还有其它几个方法：

- `contains()`
- `copy()`
- `length()`
- `valueOf()`

有关这些方法的详细信息，请参阅 [ActionScript 3.0 语言和组件参考](#)。

对于只包含一个 XML 元素的 `XMMList` 对象，可以使用 XML 类的所有属性和方法，因为包含一个 XML 元素的 `XMMList` 被视为等同于 XML 对象。例如，在下面的代码中，因为 `doc.div` 是包含一个元素的 `XMMList` 对象，所以可以使用 XML 类的 `appendChild()` 方法：

```
var doc:XML =
<body>
    <div>
        <p>Hello</p>
    </div>
</body>;
doc.div.appendChild(<p>World</p>);
```

有关 XML 属性和方法的列表，请参阅 第 192 页的“[XML 对象](#)”。

初始化 XML 变量

可将 XML 文本赋予 XML 对象，如下所示：

```
var myXML:XML =
<order>
    <item id='1'>
        <menuName>burger</menuName>
        <price>3.95</price>
    </item>
    <item id='2'>
        <menuName>fries</menuName>
        <price>1.45</price>
    </item>
</order>
```

如下面的代码片断所示，还可以使用 `new` 构造函数从包含 XML 数据的字符串创建 XML 对象的实例：

```
var str:String = "<order><item id='1'><menuName>burger</menuName>" +
    "<price>3.95</price></item></order>";
var myXML:XML = new XML(str);
```

如果字符串中的 XML 数据格式有误（例如缺少结束标签），则会出现运行时错误。

还可以将数据按引用（从其它变量）传递到 XML 对象，如下面的示例所示：

```
var tagname:String = "item";
var attributename:String = "id";
var attributevalue:String = "5";
var content:String = "Chicken";
var x:XML = <{tagname} {attributename}={attributevalue}>{content}</{tagname}>;
trace(x.toXMLString())
// Output: <item id="5">Chicken</item>
```

要从 URL 加载 XML 数据，请使用 `URLLoader` 类，如下面的示例所示：

```

import flash.events.Event;
import flash.net.URLLoader;
import flash.net.URLRequest;

var externalXML:XML;
var loader:URLLoader = new URLLoader();
var request:URLRequest = new URLRequest("xmlFile.xml");
loader.load(request);
loader.addEventListener(Event.COMPLETE, onComplete);

function onComplete(event:Event):void
{
    var loader:URLLoader = event.target as URLLoader;
    if (loader != null)
    {
        externalXML = new XML(loader.data);
        trace(externalXML.toXMLString());
    }
    else
    {
        trace("loader is not a URLLoader!");
    }
}

```

要从套接字连接读取 XML 数据，请使用 **XMLSocket** 类。有关详细信息，请参阅《ActionScript 3.0 语言和组件参考》中的 [XMLSocket 类](#)。

组合和变换 XML 对象

使用 **prependChild()** 方法或 **appendChild()** 方法可在 XML 对象属性列表的开头或结尾添加属性，如下面的示例所示：

```

var x1:XML = <p>Line 1</p>
var x2:XML = <p>Line 2</p>
var x:XML = <body></body>
x = x.appendChild(x1);
x = x.appendChild(x2);
x = x.prependChild(<p>Line 0</p>);
// x == <body><p>Line 0</p><p>Line 1</p><p>Line 2</p></body>

```

使用 **insertChildBefore()** 方法或 **insertChildAfter()** 方法在指定属性之前或之后添加属性，如下所示：

```

var x:XML =
<body>
<p>Paragraph 1</p>
<p>Paragraph 2</p>
</body>
var newNode:XML = <p>Paragraph 1.5</p>
x = x.insertChildAfter(x.p[0], newNode)
x = x.insertChildBefore(x.p[2], <p>Paragraph 1.75</p>)

```

如下面的示例所示，还可以使用大括号运算符 ({ 和 }) 在构造 XML 对象时按引用（从其它变量）传递数据：

```

var ids:Array = [121, 122, 123];
var names:Array = [["Murphy", "Pat"], ["Thibaut", "Jean"], ["Smith", "Vijay"]];
var x:XML = new XML("<employeeList></employeeList>");

for (var i:int = 0; i < 3; i++)
{
    var newnode:XML = new XML();
    newnode =
        <employee id={ids[i]}>
            <last>{names[i][0]}</last>
            <first>{names[i][1]}</first>
        </employee>

    x = x.appendChild(newnode)
}

```

可以使用 = 运算符将属性 (property) 和属性 (attribute) 赋予 XML 对象，如下所示：

```

var x:XML =
<employee>
    <lastname>Smith</lastname>
</employee>
x.firstname = "Jean";
x.@id = "239";

```

这将对 XML 对象 x 进行如下设置：

```

<employee id="239">
    <lastname>Smith</lastname>
    <firstname>Jean</firstname>
</employee>

```

可以使用 + 和 += 运算符连接 XMLList 对象：

```

var x1:XML = <a>test1</a>
var x2:XML = <b>test2</b>
var xList:XMLList = x1 + x2;
xList += <c>test3</c>

```

这将对 XMLList 对象 xList 进行如下设置：

```

<a>test1</a>
<b>test2</b>
<c>test3</c>

```

遍历 XML 结构

XML 的一个强大功能是它能够通过文本字符的线性字符串提供复杂的嵌套数据。将数据加载到 XML 对象时，ActionScript 会分析数据并将其分层结构加载到内存（如果 XML 数据格式有误，它会发送运行时错误）。

利用 XML 和 XMLList 对象的运算符和方法可以轻松遍历 XML 数据的结构。

使用点(.) 运算符和后代存取器(..) 运算符可以访问 XML 对象的子属性。请考虑下面的 XML 对象：

```

var myXML:XML =
<order>
    <book ISBN="0942407296">
        <title>Baking Extravagant Pastries with Kumquats</title>
        <author>
            <lastName>Contino</lastName>
            <firstName>Chuck</firstName>
        </author>
        <pageCount>238</pageCount>
    </book>
    <book ISBN="0865436401">
        <title>Emu Care and Breeding</title>
        <editor>
            <lastName>Case</lastName>
            <firstName>Justin</firstName>
        </editor>
        <pageCount>115</pageCount>
    </book>
</order>

```

对象 `myXML.book` 是一个 `XMLList` 对象，它包含名为 `book` 的 `myXML` 对象的子属性。它们是两个 XML 对象，与 `myXML` 对象的两个 `book` 属性匹配。

对象 `myXML..lastName` 是一个 `XMLList` 对象，该对象包含名为 `lastName` 的所有后代属性。它们是两个 XML 对象，与 `myXML` 对象的两个 `lastName` 匹配。

`myXML.book.editor.lastName` 对象是一个 `XMLList` 对象，它包含 `myXML` 对象的名为 `book` 的子对象的名为 `editor` 的子对象的名为 `lastName` 的所有子对象：在本例中，`XMLList` 对象只包含一个 XML 对象（值为 "Case" 的 `lastName` 属性）。

访问父节点和子节点

`parent()` 方法返回 XML 对象的父项。

可以使用子级列表的序数索引值访问特定的子对象。例如，假设 XML 对象 `myXML` 有两个名为 `book` 的子属性。每个名为 `book` 的子属性都有一个与之关联的索引编号：

```
myXML.book[0]
myXML.book[1]
```

要访问特定的孙项，可为子项和孙项名称同时指定索引编号：

```
myXML.book[0].title[0]
```

不过，如果 `x.book[0]` 只有一个子项名为 `title`，则可以省略索引引用，如下所示：

```
myXML.book[0].title
```

同样，如果对象 `x` 只有一个 `book` 子对象，并且如果该子对象只有一个 `title` 对象，则可以同时省略两个索引引用，如下所示：

```
myXML.book.title
```

可以使用 `child()` 方法导航到名称基于变量或表达式的子项，如下面的示例所示：

```

var myXML:XML =
<order>
    <book>
        <title>Dictionary</title>
    </book>
</order>

var childName:String = "book";

trace(myXML.child(childName).title) // output: Dictionary

```

访问属性

使用 @ 符号（属性标识符运算符）可以访问 XML 或 XMLList 对象的属性，如下面的代码所示：

```
var employee:XML =
<employee id="6401" code="233">
<lastName>Wu</lastName>
<firstName>Erin</firstName>
</employee>;
trace(employee.@id); // 6401
```

可以一起使用 * 通配符和 @ 符号来访问 XML 或 XMLList 对象的所有属性，如下面的代码所示：

```
var employee:XML =
<employee id="6401" code="233">
<lastName>Wu</lastName>
<firstName>Erin</firstName>
</employee>;
trace(employee.*.toXMLString());
// 6401
// 233
```

可以使用 attribute() 或 attributes() 方法访问 XML 或 XMLList 对象的特定属性或所有属性，如下面的代码所示：

```
var employee:XML =
<employee id="6401" code="233">
<lastName>Wu</lastName>
<firstName>Erin</firstName>
</employee>;
trace(employee.attribute("id")); // 6401
trace(employee.attribute("*").toXMLString());
// 6401
// 233
trace(employee.attributes().toXMLString());
// 6401
// 233
```

请注意，还可以使用以下语法访问属性，如下面的示例所示：

```
employee.attribute("id")
employee["@id"]
employee.@[ "id" ]
```

其中每一个都等效于 employee.@id。但是，语法 employee.@id 是首选方法。

按属性或元素值过滤

可以使用括号运算符 — (和) — 过滤具有特定元素名称或属性值的元素。请考虑下面的 XML 对象：

```
var x:XML =
<employeeList>
<employee id="347">
<lastName>Zmed</lastName>
<firstName>Sue</firstName>
<position>Data analyst</position>
</employee>
<employee id="348">
<lastName>McGee</lastName>
<firstName>Chuck</firstName>
<position>Jr. data analyst</position>
</employee>
</employeeList>
```

以下表达式都是有效的：

- `x.employee.(lastName == "McGee")` — 这是第二个 `employee` 节点。
- `x.employee.(lastName == "McGee").firstName` — 这是第二个 `employee` 节点的 `firstName` 属性。
- `x.employee.(lastName == "McGee").@id` — 这是第二个 `employee` 节点的 `id` 属性的值。
- `x.employee.(@id == 347)` — 第一个 `employee` 节点。
- `x.employee.(@id == 347).lastName` — 这是第一个 `employee` 节点的 `lastName` 属性。
- `x.employee.(@id > 300)` — 这是具有两个 `employee` 属性的 `XMLElement`。
- `x.employee.(position.toString().search("analyst") > -1)` — 这是具有两个 `position` 属性的 `XMLElement`。

如果试图按照可能不存在的属性或元素过滤，则 Flash® Player 和 Adobe® AIR™ 将引发异常。例如，以下代码的最后一行产生一个错误，因为第二个 `p` 元素中没有 `id` 属性：

```
var doc:XML =
<body>
    <p id='123'>Hello, <b>Bob</b>.</p>
    <p>Hello.</p>
</body>;
trace(doc.p.(@id == '123'));
```

同样，以下代码的最后一行也会产生一个错误，因为第二个 `p` 元素没有 `b` 属性：

```
var doc:XML =
<body>
    <p id='123'>Hello, <b>Bob</b>.</p>
    <p>Hello.</p>
</body>;
trace(doc.p.(b == 'Bob'));
```

为了避免这些错误，可以使用 `attribute()` 和 `elements()` 方法来识别具有匹配属性或元素的属性，如下面的代码所示：

```
var doc:XML =
<body>
    <p id='123'>Hello, <b>Bob</b>.</p>
    <p>Hello.</p>
</body>;
trace(doc.p.(attribute('id') == '123'));
trace(doc.p.(elements('b') == 'Bob'));
```

还可以使用 `hasOwnProperty()` 方法，如下面的代码所示：

```
var doc:XML =
<body>
    <p id='123'>Hello, <b>Bob</b>.</p>
    <p>Hello.</p>
</body>;
trace(doc.p.(hasOwnProperty('@id') && @id == '123'));
trace(doc.p.(hasOwnProperty('b') && b == 'Bob'));
```

使用 `for..in` 和 `for each..in` 语句

ActionScript 3.0 包含 `for..in` 语句和 `for each..in` 语句，用于遍历 `XMLElement` 对象。例如，我们来看 `XML` 对象 `myXML` 和 `XMLElement` 对象 `myXML.item`。`XMLElement` 对象 `myXML.item` 由 `XML` 对象的两个 `item` 节点组成。

```
var myXML:XML =
<order>
    <item id='1' quantity='2'>
        <menuName>burger</menuName>
        <price>3.95</price>
    </item>
    <item id='2' quantity='2'>
        <menuName>fries</menuName>
        <price>1.45</price>
    </item>
</order>;
```

for.in 语句用于遍历 XMLList 中的一组属性名称:

```
var total:Number = 0;
for (var pname:String in myXML.item)
{
    total += myXML.item.@quantity[pname] * myXML.item.price[pname];
}
```

for each..in 语句用于遍历 XMLList 中的属性:

```
var total2:Number = 0;
for each (var prop:XML in myXML.item)
{
    total2 += prop.@quantity * prop.price;
}
```

使用 XML 命名空间

XML 对象（或文档）中的命名空间用于标识对象所包含的数据的类型。例如，在将 XML 数据发送和提交给使用 SOAP 消息传递协议的 Web 服务时，您要在 XML 的开始标签中声明命名空间:

```
var message:XML =
<soap:Envelope xmlns:soap="http://schemas.xmlsoap.org/soap/envelope/">
    <soap:encodingStyle="http://schemas.xmlsoap.org/soap/encoding/">
        <soap:Body xmlns:w="http://www.test.com/weather/">
            <w:getWeatherResponse>
                <w:tempurature>78</w:tempurature>
            </w:getWeatherResponse>
        </soap:Body>
    </soap:Envelope>;
```

命名空间有一个前缀 `soap` 和一个定义该命名空间的 URI `http://schemas.xmlsoap.org/soap/envelope/`。

ActionScript 3.0 包含用于处理 XML 命名空间的命名空间类。对于上一示例中的 XML 对象，可按如下方式使用命名空间类:

```
var soapNS:Namespace = message.namespace("soap");
trace(soapNS); // Output: http://schemas.xmlsoap.org/soap/envelope/

var wNS:Namespace = new Namespace("w", "http://www.test.com/weather/");
message.addNamespace(wNS);
var encodingStyle:XMLList = message.@soapNS::encodingStyle;
var body:XMLList = message.soapNS::Body;

message.soapNS::Body.wNS::GetWeatherResponse.wNS::tempurature = "78";
```

XML 类包含用于处理命名空间的以下方法: `addNamespace()`、`inScopeNamespaces()`、`localName()`、`name()`、`namespace()`、`namespaceDeclarations()`、`removeNamespace()`、`setLocalName()`、`setName()` 和 `setNamespace()`。

`default xml namespace` 指令用于为 XML 对象指定默认的命名空间。例如，在以下代码中，`x1` 和 `x2` 具有相同的默认命名空间:

```
var ns1:Namespace = new Namespace("http://www.example.com/namespaces/");
default xml namespace = ns1;
var x1:XML = <test1 />;
var x2:XML = <test2 />;
```

XML 类型转换

可以将 XML 对象和 XMLList 对象转换为字符串值。同样，也可以将字符串转换为 XML 对象和 XMLList 对象。还要记住，所有 XML 属性值、名称和文本值都是字符串。以下几节将讨论所有这些形式的 XML 类型转换。

将 XML 和 XMLList 对象转换为字符串

XML 和 XMLList 类都包含一个 `toString()` 方法和一个 `toXMLString()` 方法。`toXMLString()` 方法返回包含该 XML 对象的所有标签、属性、命名空间声明和内容的字符串。对于包含复杂内容（子元素）的 XML 对象，`toString()` 方法的作用与 `toXMLString()` 方法完全相同。对于包含简单内容的 XML 对象（只包含一个文本元素的对象），`toString()` 方法只返回该元素的文本内容，如下面的示例所示：

```
var myXML:XML =
<order>
    <item id='1' quantity='2'>
        <menuName>burger</menuName>
        <price>3.95</price>
    </item>
</order>

trace(myXML.item[0].menuName.toXMLString());
// <menuName>burger</menuName>
trace(myXML.item[0].menuName.toString());
// burger
```

如果使用 `trace()` 方法但不指定 `toString()` 或 `toXMLString()`，则默认情况下将使用 `toString()` 方法转换数据，如以下代码所示：

```
var myXML:XML =
<order>
    <item id='1' quantity='2'>
        <menuName>burger</menuName>
        <price>3.95</price>
    </item>
</order>

trace(myXML.item[0].menuName);
// burger
```

使用 `trace()` 方法调试代码时，通常都要使用 `toXMLString()` 方法，以便 `trace()` 方法输出更完整的数据。

将字符串转换为 XML 对象

可以使用 `new XML()` 构造函数从字符串创建 XML 对象，如下所示：

```
var x:XML = new XML("<a>test</a>");
```

如果试图将表示无效 XML 或格式有误的 XML 的字符串转换为 XML，则会引发运行时错误，如下所示：

```
var x:XML = new XML("<a>test"); // throws an error
```

从字符串转换属性值、名称和文本值

所有 XML 属性值、名称和文本值都是 `String` 数据类型，可能需要将它们转换为其它数据类型。例如，以下代码使用 `Number()` 函数将文本值转换为数字：

```
var myXML:XML =
    <order>
        <item>
            <price>3.95</price>
        </item>
        <item>
            <price>1.00</price>
        </item>
    </order>;

var total:XML = <total>0</total>;
myXML.appendChild(total);

for each (var item:XML in myXML.item)
{
    myXML.total.children() [0] = Number(myXML.total.children() [0])
        + Number(item.price.children() [0]);
}
trace(myXML.total); // 4.35;
```

如果这些代码不使用 `Number()` 函数，它们将把 `+` 运算符解释为字符串连接运算符，并且最后一行中的 `trace()` 方法将输出以下结果：

```
01.003.95
```

读取外部 XML 文档

可以使用 `URLLoader` 类从 URL 加载 XML 数据。若要在应用程序中使用以下代码，请将示例中的 `XML_URL` 值替换为有效的 URL：

```
var myXML:XML = new XML();
var XML_URL:String = "http://www.example.com/Sample3.xml";
var myXMLURL:URLRequest = new URLRequest(XML_URL);
var myLoader:URLLoader = new URLLoader(myXMLURL);
myLoader.addEventListener("complete", xmlLoaded);

function xmlLoaded(event:Event):void
{
    myXML = XML(myLoader.data);
    trace("Data loaded.");
}
```

还可以使用 `XMLSocket` 类设置与服务器的异步 XML 套接字连接。有关详细信息，请参阅 [ActionScript 3.0 语言和组件参考](#)。

示例：从 Internet 加载 RSS 数据

`RSSViewer` 范例应用程序说明了在 ActionScript 中处理 XML 的一些功能，其中包括：

- 使用 XML 方法以 RSS 供给的形式遍历 XML 数据。
- 使用 XML 方法以 HTML 的形式组合 XML 数据，以便在文本字段中使用。

RSS 格式被广泛用于通过 XML 收集新闻。简单的 RSS 数据文件可能如下所示：

```
<?xml version="1.0" encoding="UTF-8" ?>
<rss version="2.0" xmlns:dc="http://purl.org/dc/elements/1.1/">
<channel>
    <title>Alaska - Weather</title>
    <link>http://www.nws.noaa.gov/alerts/ak.html</link>
    <description>Alaska - Watches, Warnings and Advisories</description>

    <item>
        <title>
            Short Term Forecast - Taiya Inlet, Klondike Highway (Alaska)
        </title>
        <link>
            http://www.nws.noaa.gov/alerts/ak.html#A18.AJKNK.1900
        </link>
        <description>
            Short Term Forecast Issued At: 2005-04-11T19:00:00
            Expired At: 2005-04-12T01:00:00 Issuing Weather Forecast Office
            Homepage: http://pajk.arh.noaa.gov
        </description>
    </item>
    <item>
        <title>
            Short Term Forecast - Haines Borough (Alaska)
        </title>
        <link>
            http://www.nws.noaa.gov/alerts/ak.html#AKZ019.AJKNOWAJK.190000
        </link>
        <description>
            Short Term Forecast Issued At: 2005-04-11T19:00:00
            Expired At: 2005-04-12T01:00:00 Issuing Weather Forecast Office
            Homepage: http://pajk.arh.noaa.gov
        </description>
    </item>
</channel>
</rss>
```

SimpleRSS 应用程序从 Internet 上读取 RSS 数据，分析标题、链接和描述的数据，并返回这些数据。SimpleRSSUI 类提供了相应的用户界面，并会调用 SimpleRSS 类，从而执行所有 XML 处理。

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。RSSViewer 应用程序文件位于 Samples/RSSViewer 文件夹中。该应用程序包含以下文件：

文件	描述
RSSViewer.mxml 或 RSSViewer.fla	Flash 或 Flex 中的主应用程序文件（分别为 FLA 和 MXML）。
com/example/programmingas3/rssViewer/RSSParser.as	一个类，包含的方法可以使用 E4X 遍历 RSS (XML) 数据并生成相应的 HTML 表示形式。
RSSData/ak.rss	一个 RSS 范例文件。该应用程序被设置为从 Web 上由 Adobe 托管的 Flex RSS 供给处读取 RSS 数据。不过，您也可以轻松更改该应用程序，使之从此文档读取 RSS 数据，此文档所用的架构与 Flex RSS 供给的架构略有不同。

读取和分析 XML 数据

RSSParser 类包含一个 `xmlLoaded()` 方法，该方法可将输入 RSS 数据（存储在 `rssXML` 变量中）转换为包含 HTML 格式的输出 (`rssOutput`) 的字符串。

如果源 RSS 数据包含默认的命名空间，代码会在此方法的开头附近设置默认的 XML 命名空间：

```
if (rssXML.namespace("") != undefined)
{
    default xml namespace = rssXML.namespace("");
}
```

下面的几行代码循环访问源 XML 数据的内容，以检查名为 `item` 的各个后代属性：

```
for each (var item:XML in rssXML..item)
{
    var itemTitle:String = item.title.toString();
    var itemDescription:String = item.description.toString();
    var itemLink:String = item.link.toString();
    outXML += buildItemHTML(itemTitle,
                            itemDescription,
                            itemLink);
}
```

前三行代码只是设置字符串变量，以表示 XML 数据的 `item` 属性的标题、描述和链接属性。下一行随后调用 `buildItemHTML()` 方法，以 `XMLElement` 对象形式获取 HTML 数据，并以三个新的字符串变量作为参数。

组合 XMLElement 数据

HTML 数据（`XMLElement` 对象）具有如下形式：

```
<b>itemTitle</b>
<p>
    itemDescription
    <br />
    <a href="link">
        <font color="#008000">More...</font>
    </a>
</p>
```

方法的第一行清除默认的 XML 命名空间：

```
default xml namespace = new Namespace();
default xml namespace 指令具有函数块级作用域。这意味着此声明的作用域是 buildItemHTML() 方法。
```

下面的几行代码基于传递给该函数的字符串参数组合 `XMLElement`：

```
var body:XMLElement = new XMLElement();
body += new XML("<b>" + itemTitle + "</b>");
var p:XML = new XML("<p>" + itemDescription + "</p>");

var link:XML = <a></a>;
link.@href = itemLink; // <link href="itemLinkString"></link>
link.font.@color = "#008000";
// <font color="#008000"></font></a>
// 0x008000 = green
link.font = "More...";

p.appendChild(<br/>);
p.appendChild(link);
body += p;
```

此 `XMLElement` 对象表示适用于 ActionScript HTML 文本字段的字符串数据。

xmlLoaded() 方法使用 buildItemHTML() 方法的返回值并将其转换为字符串：

```
XML.prettyPrinting = false;  
rssOutput = outXML.toXMLString();
```

提取 RSS 供给的标题并发送自定义事件

xmlLoaded() 方法基于源 RSS XML 数据中信息设置 rssTitle 字符串变量：

```
rssTitle = rssXML.channel.title.toString();
```

最后， xmlLoaded() 方法生成一个事件，通知应用程序数据已分析并且可用：

```
dataWritten = new Event("dataWritten", true);
```

第 12 章：处理事件

利用事件处理系统，程序员可以方便地响应用户输入和系统事件。ActionScript 3.0 事件模型不仅方便，而且符合标准，它与 Adobe® Flash® Player 和 Adobe® AIR™ 显示列表完美集成在一起。新的事件模型基于文档对象模型 (DOM) 第 3 级事件规范，是业界标准的事件处理体系结构，为 ActionScript 程序员提供了强大而直观的事件处理工具。

本章分为五部分。前两部分提供有关 ActionScript 中的事件处理的背景信息。最后三部分介绍了支持该事件模型的主要概念：事件流、事件对象和事件侦听器。ActionScript 3.0 事件处理系统与显示列表密切交互。本章假定您对显示列表有基本的了解。有关详细信息，请参阅第 224 页的“[显示编程](#)”。

事件处理基础知识

事件处理简介

您可以将事件视为 SWF 文件中发生的程序员感兴趣的任何类型的事件。例如，大多数 SWF 文件都支持某些类型的用户交互，无论是像响应鼠标单击这样简单的用户交互，还是像接受和处理表单中输入的数据这样复杂的用户交互。与 SWF 文件进行的任何此类用户交互都可以视为事件。也可能会在没有任何直接用户交互的情况下发生事件，例如，从服务器加载完数据或者连接的摄像头变为活动状态时。

在 ActionScript 3.0 中，每个事件都由一个事件对象表示。事件对象是 Event 类或其某个子类的实例。事件对象不但存储有关特定事件的信息，还包含便于操作事件对象的方法。例如，当 Flash Player 或 AIR 检测到鼠标单击时，它会创建一个事件对象 (MouseEvent 类的实例) 以表示该特定鼠标单击事件。

创建事件对象之后，Flash Player 或 AIR 即“调度”该事件对象，这意味着将该事件对象传递给作为事件目标的对象。作为被调度事件对象的目标的对象称为“事件目标”。例如，当连接的摄像头变为活动状态时，Flash Player 会向事件目标直接调度一个事件对象，此时，该事件对象就是代表摄像头的对象。但是，如果事件目标在显示列表中，则在显示列表层次结构中将事件对象向下传递，直到到达事件目标为止。在某些情况下，该事件对象随后会沿着相同路线在显示列表层次结构中向上“冒泡”回去。显示列表层次结构中的这种遍历行为称为“事件流”。

您可以使用事件侦听器“侦听”代码中的事件对象。“事件侦听器”是您编写的用于响应特定事件的函数或方法。若要确保您的程序响应事件，必须将事件侦听器添加到事件目标，或添加到作为事件对象事件流的一部分的任何显示列表对象。

无论何时编写事件侦听器代码，该代码都会采用以下基本结构（以粗体显示的元素是占位符，您将针对具体情况对其进行填写）：

```
function eventResponse(eventObject:EventType):void
{
    // Actions performed in response to the event go here.
}

eventTarget.addEventListener(EventType.EVENT_NAME, eventResponse);
```

此代码完成两项任务。首先，它定义一个函数，这是指定为响应事件而执行的动作的方法。接下来，调用源对象的 addEventListener() 方法，实际上就是为指定事件“订阅”该函数，以便当该事件发生时，执行该函数的操作。当事件实际发生时，事件目标将检查其注册为事件侦听器的所有函数和方法的列表。然后，它依次调用每个对象，以将事件对象作为参数进行传递。

您需要在此代码中更改四项内容以创建自己的事件侦听器。第一，必须将函数名称更改为要使用的名称（必须在两个位置更改此内容，代码将在此处显示 **eventResponse**）。第二，必须为要侦听的事件（代码中的 **EventType**）所调度的事件对象指定相应的类名称，并且必须为特定事件（列表中的 **EVENT_NAME**）指定相应的常量。第三，必须针对调度事件（此代码中的 **eventTarget**）的对象调用 addEventListener() 方法。您可以选择更改用作函数参数（此代码中的 **eventObject**）的变量的名称。

常见事件处理任务

下面是常见的事件处理任务，本章将介绍其中的每项任务：

- 编写代码以响应事件
- 阻止代码响应事件
- 处理事件对象
- 处理事件流：
 - 识别事件流信息
 - 停止事件流
 - 禁止默认行为
- 从类中调度事件
- 创建自定义事件类型

重要概念和术语

以下参考列表包含将会在本章中遇到的重要术语：

- 默认行为：某些事件包含通常与事件一起发生的行为（称为默认行为）。例如，当用户在文本字段中键入文本时，将引发文本输入事件。该事件的默认行为是实际显示在文本字段中键入的字符，但您可以覆盖该默认行为（如果由于某种原因，您不希望显示键入的字符）。
- 调度：通知事件侦听器发生了事件。
- 事件：对象可以通知其它对象它所发生的情况。
- 事件流：如果显示列表中的对象（屏幕上显示的对象）发生事件，则会向包含该对象的所有对象通知此事件，并依次通知其事件侦听器。此过程从舞台开始，并在显示列表中一直进行到发生事件的实际对象，然后再返回到舞台。此过程称为事件流。
- 事件对象：此对象包含发生的特定事件的相关信息，当调度事件时，此信息将被发送到所有侦听器。
- 事件目标：实际调度事件的对象。例如，如果用户单击位于 **Sprite**（位于舞台内）内的按钮，则所有这些对象将调度事件，但事件目标是指实际发生事件的对象，此处指单击的按钮。
- 侦听器：对象或在对象中注册其自身的函数，用于指示发生特定事件时应通知它。

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。本章中的几乎所有代码清单都包括一个 `trace()` 函数调用，用于测试代码的结果。要测试本章中的代码清单，请执行以下操作：

- 1 使用 Flash 创作工具创建空白文档
- 2 在时间轴上选择一个关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 通过“控制”>“测试影片”运行程序。

您将在“输出”面板中看到该代码清单的 `trace()` 函数的结果。

某些代码清单更为复杂一些，并且编写为类的形式。要测试这些示例，请执行以下操作：

- 1 使用 Flash 创作工具创建一个空白文档并将该文档保存到您的计算机上。

- 2 创建一个新的 ActionScript 文件，并将该文件保存到第 1 步中创建的文档所在的目录中。文件名应与代码清单中的类的名称一致。例如，如果代码清单定义一个名为 EventTest 的类，则使用名称 EventTest.as 来保存 ActionScript 文件。
- 3 将代码清单复制到 ActionScript 文件中并保存该文件。
- 4 在文档中，单击舞台或工作区的空白部分，以激活文档的“属性”检查器。
- 5 在“属性”检查器的“文档类”字段中，输入您从文本中复制的 ActionScript 类的名称。
- 6 使用“控制”>“测试影片”运行程序

您将在“输出”面板中看到该示例的结果。

用于测试示例代码清单的这些方法在第 29 页的“[测试本章内的示例代码清单](#)”中有更详细的介绍。

ActionScript 3.0 事件处理与早期版本事件处理的不同之处

ActionScript 3.0 中的事件处理与早期 ActionScript 版本中的事件处理之间的一个最显著的区别是：在 ActionScript 3.0 中，只有一个事件处理系统，而在早期的 ActionScript 版本中，则有几个不同的事件处理系统。本部分先概述早期 ActionScript 版本中的事件处理的工作原理，然后讨论 ActionScript 3.0 中的事件处理的变化情况。

早期 ActionScript 版本中的事件处理

ActionScript 3.0 之前的 ActionScript 版本中提供许多不同的方法来处理事件：

- on() 事件处理函数，可以直接放在 Button 和 MovieClip 实例上
- onClipEvent() 处理函数，可以直接放在 MovieClip 实例上
- 回调函数属性，例如 XML.onload 和 Camera.onActivity
- 使用 addListener() 方法注册的事件侦听器
- 部分实现了 DOM 事件模型的 UIEventDispatcher 类。

其中的每一种机制都有其自己的若干优点和局限性。on() 和 onClipEvent() 处理函数易于使用，但使随后对项目的维护变得较为困难，因为很难查找直接放在按钮和影片剪辑上的代码。回调函数也很容易实现，但对于任何指定事件，仅限于使用一个回调函数。事件侦听器较难实现：它们不但要求创建侦听器对象和函数，而且要求向生成事件的对象注册侦听器。这虽然增加了开销，但您可以创建若干侦听器对象，并针对同一个事件注册这些对象。

对 ActionScript 2.0 组件的开发形成了另一个事件模型。该新模型包含在 UIEventDispatcher 类中，并且基于 DOM 事件规范的子集。熟悉组件事件处理的开发人员将会发现过渡到新的 ActionScript 3.0 事件模型相对来说较为容易。

遗憾的是，各个事件模型使用的语法以不同的方式相互重叠，并且在其它方面各自不同。例如，在 ActionScript 2.0 中，某些属性（例如 TextField.onChanged）可用作回调函数或事件侦听器。但是，根据您是否在使用支持侦听器或六个类之一的 UIEventDispatcher 类，用于注册侦听器对象的语法有所不同。对于 Key、Mouse、MovieClipLoader、Selection、Stage 和 TextField 类；请使用 addListener() 方法，但对于组件事件处理，请使用名为 addEventListener() 的方法。

不同事件处理模型所导致的另一个复杂性是：根据所使用的机制的不同，事件处理函数的范围大不相同。也就是说，关键字 this 的含义在各个事件处理系统中并不一致。

ActionScript 3.0 中的事件处理

ActionScript 3.0 引入了单一事件处理模型，以替代以前各语言版本中存在的众多不同的事件处理机制。该新事件模型基于文档对象模型 (DOM) 第 3 级事件规范。虽然 SWF 文件格式并不专门遵循文档对象模型标准，但显示列表和 DOM 结构之间存在的相似性足以使 DOM 事件模型的实现成为可能。显示列表中的对象类似于 DOM 层次结构中的节点，在本讨论中，术语“显示列表对象”和“节点”可互换使用。

Flash Player 和 AIR 实现的 DOM 事件模型包括一个名为“默认行为”的概念。“默认行为”是 Flash Player 或 AIR 作为特定事件的正常后果而执行的操作。

默认行为

开发人员通常负责编写响应事件的代码。但在某些情况下，行为通常与某一事件关联，使得 Flash Player 或 AIR 会自动执行该行为，除非开发人员添加了取消该行为的代码。由于 Flash Player 或 AIR 会自动表现该行为，因此这类行为称为默认行为。

例如，当用户在 `TextField` 对象中输入文本时，普遍期待文本显示在该 `TextField` 对象中，因此该行为被内置到 Flash Player 和 AIR 中。如果您不希望该默认行为发生，可以使用新的事件处理系统来取消它。当用户在 `TextField` 对象中输入文本时，Flash Player 或 AIR 会创建 `TextEvent` 类的实例以表示该用户输入。若要阻止 Flash Player 或 AIR 显示 `TextField` 对象中的文本，必须访问该特定 `TextEvent` 实例并调用该实例的 `preventDefault()` 方法。

并非所有默认行为都可以阻止。例如，当用户双击 `TextField` 对象中的单词时，Flash Player 和 AIR 会生成一个 `MouseEvent` 对象。无法阻止的默认行为是：鼠标点击的单词突出显示。

许多类型的事件对象没有关联的默认行为。例如，当建立网络连接时，Flash Player 调度一个连接事件对象，但没有与该对象关联的默认行为。`Event` 类及其子类的 API 文档列出了每一类型的事件，并说明所有关联的默认行为，以及是否可以阻止该行为。

默认行为仅与由 Flash Player 或 AIR 所调度的事件对象关联，但通过 ActionScript 以编程方式调度的事件对象则不存在默认行为。了解这一点很重要。例如，可以使用 `EventDispatcher` 类的方法来调度类型为 `textInput` 的事件对象，但该事件对象没有关联的默认行为。也就是说，Flash Player 和 AIR 不会因为您以编程方式调度了 `textInput` 事件而在 `TextField` 对象中显示字符。

ActionScript 3.0 中事件监听器的新增功能

对于使用过 ActionScript 2.0 `addListener()` 方法的开发人员来说，指出 ActionScript 2.0 事件监听器模型和 ActionScript 3.0 事件模型之间的差别可能会有所帮助。下表说明两个事件模型之间的几个主要差别：

- 若要在 ActionScript 2.0 中添加事件监听器，请在某些情况下使用 `addListener()`，在其它情况下使用 `addEventListener()`；而在 ActionScript 3.0 中，则始终使用 `addEventListener()`。
- ActionScript 2.0 中没有事件流，这意味着，只能对广播事件的对象调用 `addListener()` 方法；而在 ActionScript 3.0 中，可以对属于事件流一部分的任何对象调用 `addEventListener()` 方法。
- 在 ActionScript 2.0 中，事件监听器可以是函数、方法或对象，而在 ActionScript 3.0 中，只有函数或方法可以是事件监听器。

事件流


只要发生事件，Flash Player 或 AIR 就会调度事件对象。如果事件目标不在显示列表中，则 Flash Player 或 AIR 将事件对象直接调度到事件目标。例如，Flash Player 将 `progress` 事件对象直接调度到 `URLStream` 对象。但是，如果事件目标在显示列表中，则 Flash Player 将事件对象调度到显示列表，事件对象将在显示列表中穿行，直到到达事件目标。

“事件流”说明事件对象如何在显示列表中穿行。显示列表以一种可以描述为树的层次结构形式进行组织。位于显示列表层次结构顶部的是舞台，它是一种特殊的显示对象容器，用作显示列表的根。舞台由 `flash.display.Stage` 类表示，且只能通过显示对象访问。每个显示对象都有一个名为 `stage` 的属性，该属性表示应用程序的舞台。


当 Flash Player 或 AIR 为显示列表相关的事件调度事件对象时，该事件对象将进行一次从舞台到“目标节点”的往返行程。DOM 事件规范将目标节点定义为代表事件目标的节点。也就是说，目标节点是发生了事件的显示列表对象。例如，如果用户单击名为 `child1` 的显示列表对象，Flash Player 或 AIR 将使用 `child1` 作为目标节点来调度事件对象。

从概念上来说，事件流分为三部分。第一部分称为捕获阶段，该阶段包括从舞台到目标节点的父节点范围内的所有节点。第二部分称为目标阶段，该阶段仅包括目标节点。第三部分称为冒泡阶段。冒泡阶段包括从目标节点的父节点返回到舞台的行程中遇到的节点。

如果您将显示列表想像为一个垂直的层次结构，其中舞台位于顶层（如下图显示），那么这些阶段的名称就更容易理解了：



如果用户单击 Child1 Node，Flash Player 或 AIR 将向事件流调度一个事件对象。如下面的图像所示，对象的行程从舞台开始，向下移到父节点，然后移到 Child1 节点，再“冒泡”返回到舞台（即在行程中重新经过父节点，再返回到舞台）。



在此示例中，捕获阶段在首次向下行程中包括舞台和父节点。目标阶段包括在 Child1 花费的时间。冒泡阶段包括在向上返回到根节点的行程中遇到的父节点和舞台。

事件流使现在的事件处理系统比 ActionScript 程序员以前使用的事件处理系统功能更为强大。早期版本的 ActionScript 中没有事件流，这意味着事件侦听器只能添加到生成事件的对象。在 ActionScript 3.0 中，您不但可以将事件侦听器添加到目标节点，还可以将它们添加到事件流中的任何节点。

当用户界面组件包含多个对象时，沿事件流添加事件侦听器的功能十分有用。例如，按钮对象通常包含一个用作按钮标签的文本对象。如果无法将侦听器添加到事件流，您将必须将侦听器添加到按钮对象和文本对象，以确保您收到有关在按钮上任何位置发生的单击事件的通知。而事件流的存在则使您可以将一个事件侦听器放在按钮对象上，以处理文本对象上发生的单击事件或按钮对象上未被文本对象遮住的区域上发生的单击事件。

不过，并非每个事件对象都参与事件流的所有三个阶段。某些类型的事件（例如 `enterFrame` 和 `init` 类型的事件）会直接调度到目标节点，并不参与捕获阶段和冒泡阶段。其它事件可能以不在显示列表中的对象为目标，例如调度到 `Socket` 类的实例的事件。这些事件对象也将直接流至目标对象，而不参与捕获和冒泡阶段。

要查明特定事件类型的行为，可以查看 API 文档或检查事件对象的属性。下面的部分介绍了如何检查事件对象的属性。

事件对象

在新的事件处理系统中，事件对象有两个主要用途。首先，事件对象通过将特定事件的信息存储在一组属性中，来代表实际事件。第二，事件对象包含一组方法，可用于操作事件对象和影响事件处理系统的行为。

为方便对这些属性和方法的访问，Flash Player API 定义了一个 `Event` 类，作为所有事件对象的基类。`Event` 类定义一组基本的适用于所有事件对象的属性和方法。

本部分首先讨论 `Event` 类属性，然后介绍 `Event` 类方法，最后说明 `Event` 类的子类存在的意义。

了解 Event 类属性

Event 类定义许多只读属性和常数，以提供有关事件对象的重要信息。以下内容尤其重要：

- 事件对象类型由常量表示，并存储在 `Event.type` 属性中。
- 事件的默认行为是否可以被阻止由布尔值表示，并存储在 `Event.cancelable` 属性中。
- 事件流信息包含在其余属性中。

事件对象类型

每个事件对象都有关联的事件类型。数据类型以字符串值的形式存储在 `Event.type` 属性中。知道事件对象的类型是非常有用的，这样您的代码就可以区分不同类型的事件。例如，下面的代码指定 `clickHandler()` 倾听器函数应响应传递给 `myDisplayObject` 的任何鼠标单击事件对象。

```
myDisplayObject.addEventListener(MouseEvent.CLICK, clickHandler);
```

大约有 20 多种事件类型与 Event 类自身关联并由 Event 类常数表示，其中某些数据类型显示在摘自 Event 类定义的以下代码中：

```
package flash.events
{
    public class Event
    {
        // class constants
        public static const ACTIVATE:String = "activate";
        public static const ADDED:String= "added";
        // remaining constants omitted for brevity
    }
}
```

这些常数提供了引用特定事件类型的简便方法。您应使用这些常数而不是它们所代表的字符串。如果您的代码中拼错了某个常数名称，编译器将捕获到该错误，但如果您改为使用字符串，则编译时可能不会出现拼写错误，这可能导致难以调试的意外行为。例如，添加事件倾听器时，使用以下代码：

```
myDisplayObject.addEventListener(MouseEvent.CLICK, clickHandler);
```

而不是使用：

```
myDisplayObject.addEventListener("click", clickHandler);
```

默认行为信息

代码可通过访问 `cancelable` 属性来检查是否可以阻止任何给定事件对象的默认行为。`cancelable` 属性包含一个布尔值，用于指示是否可以阻止默认行为。您可以使用 `preventDefault()` 方法阻止或取消与少量事件关联的默认行为。有关详细信息，请参阅第 213 页的“[了解 Event 类方法](#)”下的“取消默认事件行为”。

事件流信息

其余 Event 类属性包含有关事件对象及其与事件流的关系的重要信息，如以下列表所述：

- `bubbles` 属性包含有关事件流中事件对象参与的部分的信息。
- `eventPhase` 属性指示事件流中的当前阶段。
- `target` 属性存储对事件目标的引用。
- `currentTarget` 属性存储对当前正在处理事件对象的显示列表对象的引用。

bubbles 属性

如果事件对象参与事件流的冒泡阶段，则将该事件称为“冒泡”，这指的是从目标节点将事件对象往回传递，经过目标节点的父节点，直到到达舞台。Event.bubbles 属性存储一个布尔值，用于指示事件对象是否参与冒泡阶段。由于冒泡的所有事件还参与捕获和目标阶段，因此这些事件参与事件流的所有三个阶段。如果值为 `true`，则事件对象参与所有三个阶段。如果值为 `false`，则事件对象不参与冒泡阶段。

eventPhase 属性

您可以通过调查任何事件对象的 `eventPhase` 属性来确定事件阶段。`eventPhase` 属性包含一个无符号整数值，该值代表三个事件流阶段中的一个阶段。Flash Player API 定义了单独的 `EventPhase` 类，该类包含三个对应于三个无符号整数值的常量，如以下摘录代码中所示：

```
package flash.events
{
    public final class EventPhase
    {
        public static const CAPTURING_PHASE:uint = 1;
        public static const AT_TARGET:uint = 2;
        public static const BUBBLING_PHASE:uint = 3;
    }
}
```

这些常量对应于 `eventPhase` 属性的三个有效值。使用这些常数可以使您的代码可读性更好。例如，如果要确保仅当事件目标在目标阶段中时才调用名为 `myFunc()` 的函数，您可以使用以下代码来测试此条件：

```
if (event.eventPhase == EventPhase.AT_TARGET)
{
    myFunc();
}
```

target 属性

`target` 属性包含对作为事件目标的对象的引用。在某些情况下，这很简单，例如当麦克风变为活动状态时，事件对象的目标是 `Microphone` 对象。但是，如果目标在显示列表中，则必须考虑显示列表层次结构。例如，如果用户在包括重叠的显示列表对象的某一点输入一个鼠标单击，则 Flash Player 和 AIR 始终会选择距离舞台层次最深的对象作为事件目标。

对于复杂的 SWF 文件，特别是那些通常使用更小的子对象来修饰按钮的 SWF 文件，`target` 属性可能并不常用，因为它通常指向按钮的子对象，而不是按钮。在这些情况下，常见的做法是将事件侦听器添加到按钮并使用 `currentTarget` 属性，因为该属性指向按钮，而 `target` 属性可能指向按钮的子对象。

currentTarget 属性

`currentTarget` 属性包含对当前正在处理事件对象的对象的引用。您并不知道哪个节点当前正在处理您要检查的事件对象，虽然这似乎很奇怪，但请记住，您可以向该事件对象的事件流中的任何显示对象添加侦听器函数，并且可以将侦听器函数放在任何位置。而且，可以将相同的侦听器函数添加到不同的显示对象。随着项目大小和复杂性的增加，`currentTarget` 属性会变得越来越有用。

了解 Event 类方法

共有三种类别的 Event 类方法：

- 实用程序方法：可以创建事件对象的副本或将其转换为字符串
- 事件流方法：用于从事件流中删除事件对象
- 默认行为方法：可阻止默认行为或检查是否已阻止默认行为

Event 流实用程序方法

Event 类中有两个实用程序方法。`clone()` 方法用于创建事件对象的副本。`toString()` 方法用于生成事件对象属性的字符串表示形式以及它们的值。这两个方法都由事件模型系统在内部使用，但对开发人员公开以用于一般用途。

对于创建 Event 类的子类的高级开发人员来说，必须覆盖和实现两个实用程序方法的版本，以确保事件子类正常使用。

停止事件流

可以调用 `Event.stopPropagation()` 方法或 `Event.stopImmediatePropagation()` 方法来阻止在事件流中继续执行事件对象。这两种方法几乎相同，唯一的不同之处在于是否允许执行当前节点的其它事件监听器：

- `Event.stopPropagation()` 方法可阻止事件对象移到下一个节点，但只有在允许执行当前节点上的任何其它事件监听器之后才起作用。
- `Event.stopImmediatePropagation()` 方法也可阻止事件对象移到下一个节点，但不允许执行当前节点上的任何其它事件监听器。

调用其中任何一个方法对是否发生与事件关联的默认行为没有影响。使用 Event 类的默认行为方法可以阻止默认行为。

取消默认事件行为

与取消默认行为有关的两个方法是 `preventDefault()` 方法和 `isDefaultPrevented()` 方法。调用 `preventDefault()` 方法可取消与事件关联的默认行为。若要查看是否已针对事件对象调用了 `preventDefault()`，请调用 `isDefaultPrevented()` 方法；如果已经调用，该方法将返回值 `true`，否则返回值 `false`。

`preventDefault()` 方法仅在可以取消事件的默认行为时才起作用。可通过参考该事件类型的 API 文档或使用 ActionScript 检查事件对象的 `cancelable` 属性来确定是否属于这种情况。

取消默认行为对事件对象通过事件流的进度没有影响。使用 Event 类的事件流方法可以从事件流中删除事件对象。

Event 类的子类

对于很多事件，Event 类中定义的一组公共属性已经足够了。但是，Event 类中可用的属性无法捕获其它事件具有的独特的特性。ActionScript 3.0 为这些事件定义了 Event 类的几个子类。

每个子类提供了对于该事件类别来说唯一的附加属性和事件类型。例如，与鼠标输入相关的事件具有若干独特的特性，无法被 Event 类中定义的属性捕获。MouseEvent 类添加了 10 个属性，扩展了 Event 类。这 10 个属性包含诸如鼠标事件的位置和在鼠标事件过程中是否按下了特定键等信息。

Event 子类还包含代表与子类关联的事件类型的常量。例如，MouseEvent 类定义几种鼠标事件类型的内容，包括 `click`、`doubleClick`、`mouseDown` 和 `mouseUp` 事件类型。

正如第 211 页的“[事件对象](#)”下的“Event 类实用程序方法”一节所述，创建 Event 子类时，必须覆盖 `clone()` 和 `toString()` 方法以实现特定于该子类的功能。

事件监听器

事件监听器也称为事件处理函数，是 Flash Player 和 AIR 为响应特定事件而执行的函数。添加事件监听器的过程分为两步。首先，为 Flash Player 或 AIR 创建一个为响应事件而执行的函数或类方法。这有时称为监听器函数或事件处理函数。然后，使用 `addEventListener()` 方法，在事件的目标或位于适当事件流上的任何显示列表对象中注册监听器函数。

创建侦听器函数

创建侦听器函数是 ActionScript 3.0 事件模型与 DOM 事件模型不同的一个方面。在 DOM 事件模型中，事件侦听器和侦听器函数之间有一个明显不同：即事件侦听器是实现 `EventListener` 接口的类的实例，而侦听器是该类的名为 `handleEvent()` 的方法。在 DOM 事件模型中，您注册的是包含侦听器函数的类实例，而不是实际的侦听器函数。

在 ActionScript 3.0 事件模型中，事件侦听器和侦听器函数之间没有区别。ActionScript 3.0 没有 `EventListener` 接口，侦听器函数可以在类外部定义，也可以定义为类的一部分。此外，无需将侦听器函数命名为 `handleEvent()` — 可以将它们命名为任何有效的标识符。在 ActionScript 3.0 中，您注册的是实际侦听器函数的名称。

在类外部定义的侦听器函数

以下代码创建一个显示红色正方形的简单 SWF 文件。名为 `clickHandler()` 的侦听器函数（不是类的一部分）侦听红色正方形上的鼠标单击事件。

```
package
{
    import flash.display.Sprite;

    public class ClickExample extends Sprite
    {
        public function ClickExample()
        {
            var child:ChildSprite = new ChildSprite();
            addChild(child);
        }
    }

    import flash.display.Sprite;
    import flash.events.MouseEvent;

    class ChildSprite extends Sprite
    {
        public function ChildSprite()
        {
            graphics.beginFill(0xFF0000);
            graphics.drawRect(0,0,100,100);
            graphics.endFill();
            addEventListener(MouseEvent.CLICK, clickHandler);
        }
    }

    function clickHandler(event:MouseEvent):void
    {
        trace("clickHandler detected an event of type: " + event.type);
        trace("the this keyword refers to: " + this);
    }
}
```

当用户通过单击正方形与生成的 SWF 文件交互时，Flash Player 或 AIR 生成以下跟踪输出：

```
clickHandler detected an event of type: click
the this keyword refers to: [object global]
```

请注意，事件对象作为参数传递到 `clickHandler()`。这就允许您的侦听器函数检查事件对象。在该示例中，使用事件对象的 `type` 属性来确定该事件是否为单击事件。

该示例还检查 `this` 关键字的值。在本例中，`this` 表示全局对象，这是因为函数是在任何自定义类或对象外部定义的。

定义为类方法的侦听器函数

下面的示例与前面定义 `ClickExample` 类的示例相同，只是将 `clickHandler()` 函数定义为 `ChildSprite` 类的方法：

```

package
{
    import flash.display.Sprite;

    public class ClickExample extends Sprite
    {
        public function ClickExample()
        {
            var child:ChildSprite = new ChildSprite();
            addChild(child);
        }
    }
}

import flash.display.Sprite;
import flash.events.MouseEvent;

class ChildSprite extends Sprite
{
    public function ChildSprite()
    {
        graphics.beginFill(0xFF0000);
        graphics.drawRect(0,0,100,100);
        graphics.endFill();
        addEventListener(MouseEvent.CLICK, clickHandler);
    }
    private function clickHandler(event:MouseEvent):void
    {
        trace("clickHandler detected an event of type: " + event.type);
        trace("the this keyword refers to: " + this);
    }
}

```

当用户通过单击红色正方形与生成的 SWF 文件交互时，Flash Player 或 AIR 生成以下跟踪输出：

```

clickHandler detected an event of type: click
the this keyword refers to: [object ChildSprite]

```

请注意，this 关键字引用名为 child 的 ChildSprite 实例。这是与 ActionScript 2.0 相比行为方面的一个变化。如果您使用过 ActionScript 2.0 中的组件，您可能会记得，将类方法传递给 UIEventDispatcher.addEventListener() 时，方法的作用域被绑定到广播事件的组件，而不是在其中定义侦听器方法的类。也就是说，如果在 ActionScript 2.0 中使用该技术，this 关键字将引用广播事件的组件，而不是 ChildSprite 实例。

这对于某些程序员来说是一个重要问题，因为这意味着他们无法访问包含侦听器方法的类的其它方法和属性。过去，ActionScript 2.0 程序员可以使用 mx.util.Delegate 类更改侦听器方法的作用域以解决该问题。不过，现在已不再需要这样做了，因为 ActionScript 3.0 在调用 addEventListener() 时会创建一个绑定方法。这样，this 关键字引用名为 child 的 ChildSprite 实例，且程序员可以访问 ChildSprite 类的其它方法和属性。

不应使用的事件侦听器

还有第三种技术可用于创建一个通用对象，该对象具有指向动态分配的侦听器函数的属性，但不推荐使用该技术。在此讨论这种技术是因为它在 ActionScript 2.0 中经常使用，但在 ActionScript 3.0 中不应使用。建议不要使用此项技术，因为 this 关键字将引用全局对象，而不引用您的侦听器对象。

下面的示例与前面的 ClickExample 类示例相同，只是将侦听器函数定义为名为 myListenerObj 的通用对象的一部分：

```
package
{
    import flash.display.Sprite;

    public class ClickExample extends Sprite
    {
        public function ClickExample()
        {
            var child:ChildSprite = new ChildSprite();
            addChild(child);
        }
    }
}

import flash.display.Sprite;
import flash.events.MouseEvent;

class ChildSprite extends Sprite
{
    public function ChildSprite()
    {
        graphics.beginFill(0xFF0000);
        graphics.drawRect(0,0,100,100);
        graphics.endFill();
        addEventListener(MouseEvent.CLICK, myListenerObj.clickHandler);
    }
}

var myListenerObj:Object = new Object();
myListenerObj.clickHandler = function (event:MouseEvent):void
{
    trace("clickHandler detected an event of type: " + event.type);
    trace("the this keyword refers to: " + this);
}
```

跟踪的结果将类似如下：

```
clickHandler detected an event of type: click
the this keyword refers to: [object global]
```

您可能以为 this 引用 myListenerObj，且跟踪输出应为 [object Object]，但实际上它引用的是全局对象。将动态属性名称作为参数传递给 addEventListener() 时，Flash Player 或 AIR 无法创建绑定方法。这是因为作为 listener 参数传递的只不过是侦听器函数的内存地址，Flash Player 和 AIR 无法将该内存地址与 myListenerObj 实例关联起来。

管理事件侦听器

使用 IEventDispatcher 接口的方法来管理侦听器函数。IEventDispatcher 接口是 ActionScript 3.0 版本的 DOM 事件模型的 EventTarget 接口。虽然名称 IEventDispatcher 似乎暗示着其主要用途是发送（调度）事件对象，但该类的方法实际上更多用于注册、检查和删除事件侦听器。IEventDispatcher 接口定义五个方法，如以下代码中所示：

```

package flash.events
{
    public interface IEventDispatcher
    {
        function addEventListener(eventName:String,
            listener:Object,
            useCapture:Boolean=false,
            priority:Integer=0,
            useWeakReference:Boolean=false):Boolean;

        function removeEventListener(eventName:String,
            listener:Object,
            useCapture:Boolean=false):Boolean;

        function dispatchEvent(eventObject:Event):Boolean;

        function hasEventListener(eventName:String):Boolean;
        function willTrigger(eventName:String):Boolean;
    }
}

```

Flash Player API 使用 EventDispatcher 类来实现 IEventDispatcher 接口，该类用作可以是事件目标或事件流一部分的所有类的基类。例如，DisplayObject 类继承自 EventDispatcher 类。这意味着，显示列表中的所有对象都可以访问 IEventDispatcher 接口的方法。

添加事件侦听器

`addEventListener()` 方法是 `IEventDispatcher` 接口的主要函数。使用它来注册侦听器函数。两个必需的参数是 `type` 和 `listener`。`type` 参数用于指定事件的类型。`listener` 参数用于指定发生事件时将执行的侦听器函数。`listener` 参数可以是对函数或类方法的引用。

注：指定 `listener` 参数时，不要使用括号。例如，在下面的 `addEventListener()` 方法调用中，指定 `clickHandler()` 函数时没有使用括号：

注：`addEventListener(MouseEvent.CLICK, clickHandler);`

通过使用 `addEventListener()` 方法的 `useCapture` 参数，可以控制侦听器将处于活动状态的事件流阶段。如果 `useCapture` 设置为 `true`，侦听器将在事件流的捕获阶段成为活动状态。如果 `useCapture` 设置为 `false`，侦听器将在事件流的目标阶段和冒泡阶段处于活动状态。若要在事件流的所有阶段监听某一事件，您必须调用 `addEventListener()` 两次，第一次调用时将 `useCapture` 设置为 `true`，第二次调用时将 `useCapture` 设置为 `false`。

`addEventListener()` 方法的 `priority` 参数并不是 DOM Level 3 事件模型的正式部分。ActionScript 3.0 中包括它是为了在组织事件侦听器时提供更大的灵活性。调用 `addEventListener()` 时，可以将一个整数值作为 `priority` 参数传递，以设置该事件侦听器的优先级。默认值为 0，但您可以将它设置为负整数值或正整数值。将优先执行此数字较大的事件侦听器。对于具有相同优先级的事件侦听器，则按它们的添加顺序执行，因此将优先执行较早添加的侦听器。

可以使用 `useWeakReference` 参数来指定对侦听器函数的引用是弱引用还是正常引用。通过将此参数设置为 `true`，可避免侦听器函数在不再需要时仍然存在于内存中的情况。Flash Player 和 AIR 使用一项称为“垃圾回收”的技术从内存中清除不再使用的对象。如果不存在对某个对象的引用，则该对象被视为不再使用。垃圾回收器不考虑弱引用，这意味着如果侦听器函数仅具有指向它的弱引用，则符合垃圾回收条件。

删除事件侦听器

可以使用 `removeEventListener()` 方法删除不再需要的事件侦听器。删除将不再使用的所有侦听器是个好办法。必需的参数包括 `eventName` 和 `listener` 参数，这些参数与 `addEventListener()` 方法的必需参数相同。回想一下，您可以通过调用 `addEventListener()` 两次（第一次调用时将 `useCapture` 设置为 `true`，第二次调用时将其设置为 `false`），在所有事件阶段监听事件。若要删除这两个事件侦听器，您需要调用 `removeEventListener()` 两次，第一次调用时将 `useCapture` 设置为 `true`，第二次调用时将其设置为 `false`。

发送事件

高级程序员可以使用 `dispatchEvent()` 方法将自定义事件对象调度到事件流。该方法唯一接受的参数是对事件对象的引用，此事件对象必须是 `Event` 类的实例或子类。调度后，事件对象的 `target` 属性将设置为对其调用了 `dispatchEvent()` 的对象。

检查现有的事件侦听器

`IEventDispatcher` 接口的最后两个方法提供有关是否存在事件侦听器的有用信息。如果在特定显示列表对象上发现特定事件类型的事件侦听器，`hasEventListener()` 方法将返回 `true`。如果发现特定显示列表对象的侦听器，`willTrigger()` 方法也会返回 `true`。但 `willTrigger()` 不但检查该显示对象上的侦听器，还会检查该显示列表对象在事件流所有阶段中的所有始祖上的侦听器。

没有侦听器的错误事件

ActionScript 3.0 中处理错误的主要机制是异常而不是事件，但对于异步操作（例如加载文件），异常处理不起作用。如果在这样的异步操作中发生错误，Flash Player 和 AIR 会调度一个错误事件对象。如果不为错误事件创建侦听器，Flash Player 和 AIR 的调试器版本将打开一个对话框，其中包含有关该错误的信息。例如，调试版的 Flash Player 会在应用程序试图从无效 URL 加载文件时生成以下对话框来描述错误：



大多数错误事件基于 `ErrorEvent` 类，而且同样具有一个名为 `text` 的属性，它用于存储 Flash Player 或 AIR 显示的错误消息。两个异常是 `StatusEvent` 和 `NetStatusEvent` 类。这两个类都具有一个 `level` 属性 (`StatusEvent.level` 和 `NetStatusEvent.info.level`)。当 `level` 属性的值为 "error" 时，这些事件类型被视为错误事件。

错误事件将不会导致 SWF 文件停止运行。它仅在浏览器插件和独立播放器的调试器版本上显示为对话框，在创作播放器的输出面板中显示为消息，在 Adobe Flex Builder 3 的日志文件中显示为条目。而在 Flash Player 或 AIR 的发行版中根本不会显示。

示例：Alarm Clock

Alarm Clock 示例包含一个时钟，供用户指定闹铃响起的时间，还包含一个在该时间显示的消息。Alarm Clock 示例是在第 108 页的“[处理日期和时间](#)”中的 SimpleClock 应用程序的基础上构建的。Alarm Clock 说明了在 ActionScript 3.0 中使用事件的几个方面，其中包括：

- 侦听和响应事件
- 向侦听器通知事件
- 创建自定义事件类型

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可以在 Samples/AlarmClock 文件夹中找到 Alarm Clock 应用程序文件。该应用程序包括以下文件：

文件	描述
AlarmClockApp.mxml 或 AlarmClockApp.fla	Flash 或 Flex 中的主应用程序文件（分别为 FLA 和 MXML）。
com/example/programmingas3/clock/AlarmClock.as	一个扩展 SimpleClock 类的类，添加了闹钟功能。
com/example/programmingas3/clock/AlarmEvent.as	自定义事件类（flash.events.Event 的子类），用作 AlarmClock 类的 alarm 事件的事件对象。
com/example/programmingas3/clock/AnalogClockFace.as	绘制一个圆的时钟形状以及基于时间的时针、分针和秒针（如 SimpleClock 示例中所述）。
com/example/programmingas3/clock/SimpleClock.as	具有简单走时功能的时钟界面组件（如 SimpleClock 示例中所述）。

Alarm Clock 概述

在此示例中，时钟的主要功能（包括跟踪时间和显示时钟形状）重复使用 SimpleClock 应用程序代码，相关介绍请参阅第 113 页的“[示例：简单的模拟时钟](#)”。AlarmClock 类添加了闹钟所需的功能（包括设置闹铃时间和在闹铃“响起”时显示通知），从而扩展了该示例中的 SimpleClock 类。

在发生事情时提供通知，是创建事件的目的。AlarmClock 类公开 Alarm 事件，其它对象可侦听该事件以执行所需操作。此外，AlarmClock 类使用 Timer 类的实例来确定何时触发闹铃。和 AlarmClock 类一样，Timer 类提供一个事件，用于在经过特定时间后通知其它对象（在本例中为 AlarmClock 实例）。就像大多数 ActionScript 应用程序一样，事件构成了 Alarm Clock 范例应用程序功能的重要部分。

触发闹铃

如前所述，AlarmClock 类实际提供的唯一功能与设置和触发闹铃有关。内置的 Timer 类（flash.utils.Timer）为开发人员提供了定义要在指定时间之后执行的代码的方法。AlarmClock 类使用 Timer 实例来确定何时触发闹铃。

```
import flash.events.TimerEvent;
import flash.utils.Timer;

/**
 * The Timer that will be used for the alarm.
 */
public var alarmTimer:Timer;

...
/** 
 * Instantiates a new AlarmClock of a given size.
 */
public override function initClock(faceSize:Number = 200):void
{
    super.initClock(faceSize);
    alarmTimer = new Timer(0, 1);
    alarmTimer.addEventListener(TimerEvent.TIMER, onAlarm);
}
```

AlarmClock 类中定义的 Timer 实例被命名为 alarmTimer。initClock() 方法执行 AlarmClock 实例的所需设置操作，使用 alarmTimer 变量执行两个任务。首先，使用指示 Timer 实例等待 0 毫秒且仅触发其 timer 事件一次的参数实例化变量。实例化 alarmTimer 后，代码调用变量的 addEventListener() 方法，指示它要监听该变量的 timer 事件。Timer 实例的工作方式是：在经过指定时间后调度其 timer 事件。AlarmClock 类需要了解何时调度 timer 事件，以便触发自己的闹铃。通过调用 addEventListener()，AlarmClock 代码将自身作为侦听器在 alarmTimer 中进行注册。两个参数指示 AlarmClock 类要侦听 timer 事件（由常量 TimerEvent.TIMER 指示），并且当事件发生时，应调用 AlarmClock 类的 onAlarm() 方法以响应事件。

为了实际设置闹铃，代码调用了 AlarmClock 类的 setAlarm() 方法，如下所示：

```

/**
 * Sets the time at which the alarm should go off.
 * @param hour The hour portion of the alarm time.
 * @param minutes The minutes portion of the alarm time.
 * @param message The message to display when the alarm goes off.
 * @return The time at which the alarm will go off.
 */
public function setAlarm(hour:Number = 0, minutes:Number = 0, message:String = "Alarm!"):Date
{
    this.alarmMessage = message;
    var now:Date = new Date();
    // Create this time on today's date.
    alarmTime = new Date(now.fullYear, now.month, now.date, hour, minutes);

    // Determine if the specified time has already passed today.
    if (alarmTime <= now)
    {
        alarmTime.setTime(alarmTime.time + MILLISECONDS_PER_DAY);
    }

    // Stop the alarm timer if it's currently set.
    alarmTimer.reset();
    // Calculate how many milliseconds should pass before the alarm should
    // go off (the difference between the alarm time and now) and set that
    // value as the delay for the alarm timer.
    alarmTimer.delay = Math.max(1000, alarmTime.time - now.time);
    alarmTimer.start();

    return alarmTime;
}

```

此方法执行了几项操作，包括存储闹铃消息和创建一个 `Date` 对象 (`alarmTime`)，该对象表示触发闹铃的实际时间。在该方法的最后几行中，与当前讨论最相关的操作是设置和激活了 `alarmTimer` 变量的计时器。首先，调用其 `reset()` 方法，如果计时器已运行，则将其停止并进行重置。接下来，从 `alarmTime` 变量值中减去当前时间（由 `now` 变量表示），以确定需要经过多少毫秒后才会触发闹铃。`Timer` 类并不会在某个绝对时间触发其 `timer` 事件，因此，分配给 `alarmTimer` 的 `delay` 属性的是该相对时间差。最后，调用 `start()` 方法以实际启动计时器。

一旦经过指定时间，`alarmTimer` 将调度 `timer` 事件。由于 `AlarmClock` 类已将其 `onAlarm()` 方法注册为该事件的侦听器，因此发生 `timer` 事件时，将调用 `onAlarm()`。

```

/**
 * Called when the timer event is dispatched.
 */
public function onAlarm(event:TimerEvent):void
{
    trace("Alarm!");
    var alarm:AlarmEvent = new AlarmEvent(this.alarmMessage);
    this.dispatchEvent(alarm);
}

```

注册为事件侦听器的方法必须使用适当的签名（即，方法的参数集和返回类型）来定义。若要侦听 `Timer` 类的 `timer` 事件，方法必须定义一个数据类型为 `TimerEvent` (`flash.events.TimerEvent`) 的参数，该参数是 `Event` 类的子类。当 `Timer` 实例调用其事件侦听器时，会传递一个 `TimerEvent` 实例作为事件对象。

向其它代码通知闹铃

和 Timer 类一样， AlarmClock 类提供了一个事件，以允许其它代码在闹铃响起时收到通知。对于类而言，要使用内置于 ActionScript 中的事件处理框架，必须实现 flash.events.IEventDispatcher 接口。通常，这是通过扩展 flash.events.EventDispatcher 类（提供 IEventDispatcher 的标准实现）或 EventDispatcher 的某个子类来完成的。如前所述， AlarmClock 类扩展 SimpleClock 类， SimpleClock 类扩展 Sprite 类，而 Sprite 类（通过继承链）扩展 EventDispatcher 类。所有这些意味着 AlarmClock 类已经具有内置功能以提供自己的事件。

其它代码可通过调用 AlarmClock 从 EventDispatcher 继承的 addEventListener() 方法进行注册，以获得 AlarmClock 类的 alarm 事件的通知。当 AlarmClock 实例准备通知其它代码已引发其 alarm 事件时，它会调用 dispatchEvent() 方法进行通知，该方法同样是从 EventDispatcher 继承的。

```
var alarm:AlarmEvent = new AlarmEvent(this.alarmMessage);
this.dispatchEvent(alarm);
```

这些代码行摘自 AlarmClock 类的 onAlarm() 方法（前面完整介绍过）。调用 AlarmClock 实例的 dispatchEvent() 方法，该方法接下来通知所有注册的侦听器：已触发 AlarmClock 实例的 alarm 事件。传递给 dispatchEvent() 的参数是要一直传递到侦听器方法的事件对象。在本例中，它是 AlarmEvent 类的实例，即为本示例专门创建的 Event 子类。

提供自定义 Alarm 事件

所有事件侦听器都接收一个事件对象参数，该参数提供有关要触发的特定事件的信息。在许多情况下，事件对象是 Event 类的实例。但在某些情况下，向事件侦听器提供其它信息很有用。如本章前面所述，实现该目的的一个常用方法是定义一个新类（Event 类的子类），并将该类的实例用作事件对象。在本示例中，当调度 AlarmClock 类的 alarm 事件时，会将一个 AlarmEvent 实例用作事件对象。在此介绍的 AlarmEvent 类提供有关 alarm 事件的其它信息，具体来说是闹铃消息：

```
import flash.events.Event;

/**
 * This custom Event class adds a message property to a basic Event.
 */
public class AlarmEvent extends Event
{
    /**
     * The name of the new AlarmEvent type.
     */
    public static const ALARM:String = "alarm";

    /**
     * A text message that can be passed to an event handler
     * with this event object.
     */
    public var message:String;

    /**
     *Constructor.
     *@param message The text to display when the alarm goes off.
     */
    public function AlarmEvent(message:String = "ALARM!")
    {
        super(ALARM);
        this.message = message;
    }
    ...
}
```

要创建自定义事件对象类，最好的方法是定义一个扩展 Event 类的类，如前面的示例中所示。为了补充继承的功能，AlarmEvent 类定义一个属性 message，该属性包含与事件关联的闹铃消息的文本； message 是作为 AlarmEvent 构造函数中的参数传入的。 AlarmEvent 类还定义常量 ALARM，当调用 AlarmClock 类的 addEventListener() 方法时，该常量可用于引用特定事件 (alarm)。

除了添加自定义功能外，作为 ActionScript 事件处理框架的一部分，每个 Event 子类还必须覆盖继承的 clone() 方法。Event 子类还可以选择性地覆盖继承的 toString() 方法，以便在调用 toString() 方法时返回的值中包括自定义事件的属性。

```
/**  
 * Creates and returns a copy of the current instance.  
 * @return A copy of the current instance.  
 */  
public override function clone():Event  
{  
    return new AlarmEvent(message);  
}  
  
/**  
 * Returns a String containing all the properties of the current  
 * instance.  
 * @return A string representation of the current instance.  
 */  
public override function toString():String  
{  
    return formatToString("AlarmEvent", "type", "bubbles", "cancelable", "eventPhase", "message");  
}
```

被覆盖的 clone() 方法需要返回自定义 Event 子类的新实例，并且设置了所有自定义属性以匹配当前实例。在被覆盖的 toString() 方法中，实用程序方法 formatToString()（从 Event 继承）用于提供一个字符串，包括自定义类型的名称以及所有属性的名称和值。


第 13 章：显示编程

Adobe® ActionScript® 3.0 中的显示编程可用于处理显示在 Adobe® Flash® Player 或 Adobe® AIR™ 舞台上的元素。本章介绍有关使用屏幕元素的基本概念。您将了解有关以编程方式组织可视元素的详细信息。还将了解如何为显示对象创建自己的自定义类。

显示编程的基础知识

显示编程简介

使用 ActionScript 3.0 构建的每个应用程序都有一个由显示对象构成的层次结构，这个结构称为“显示列表”，如下所示。显示列表包含应用程序中的所有可视元素。



如图所示，显示元素分为以下一个或多个组：

- 舞台

舞台是包括显示对象的基础容器。每个应用程序都有一个 **Stage** 对象，其中包含所有的屏幕显示对象。舞台是顶级容器，它位于显示列表层次结构的顶部：

每个 SWF 文件都有一个关联的 ActionScript 类，该类称为“SWF 文件的主类”。在 Flash Player 或 Adobe AIR 中打开 SWF 文件时，Flash Player 或 AIR 将调用该类的构造函数，并添加所创建的实例（始终是一种显示对象）作为 Stage 对象的子级。SWF 文件的主类始终扩展 **Sprite** 类（有关详细信息，请参阅第 229 页的“[显示列表方法的优点](#)”）。

可以通过任何 **DisplayObject** 实例的 **stage** 属性来访问舞台。有关详细信息，请参阅第 236 页的“[设置舞台属性](#)”。


- 显示对象

在 ActionScript 3.0 中，在应用程序屏幕上出现的所有元素都属于“显示对象”类型。**flash.display** 包中有一个 **DisplayObject** 类，该类是许多其它类进行扩展的基类。这些不同的类表示一些不同类型的显示对象，如矢量形状、影片剪辑和文本字段等。有关这些类的概述，请参阅第 229 页的“[显示列表方法的优点](#)”。

- 显示对象容器

显示对象容器是一些特殊类型的显示对象，这些显示对象除了有自己的可视表示形式之外，还可以包含也是显示对象的子对象。

DisplayObjectContainer 类是 **DisplayObject** 类的子类。**DisplayObjectContainer** 对象可以在其“子级列表”中包含多个显示对象。例如，下图显示一种称为 **Sprite** 的 **DisplayObjectContainer** 对象，其中包含各种显示对象：



A. **SimpleButton** 对象。此类显示对象有不同的“弹起”、“按下”和“指针经过”状态。 **B.** **Bitmap** 对象。本例中，**Bitmap** 对象是通过 Loader 对象从外部 JPEG 加载的。 **C.** **Shape** 对象。“图片帧”包含一个在 ActionScript 中绘制的圆角矩形。此 **Shape** 对象有一个应用于它的 Drop Shadow 滤镜。 **D.** 一个 **TextField** 对象。

在讨论显示对象的上下文中，**DisplayObjectContainer** 对象又称为“显示对象容器”或简称为“容器”。如前所述，舞台是显示对象容器。

尽管所有可视显示对象都从 **DisplayObject** 类继承，但每类显示对象都是 **DisplayObject** 类的一个特定子类。例如，有 **Shape** 类或 **Video** 类的构造函数，但没有 **DisplayObject** 类的构造函数。

常见的显示编程任务

由于这么多的 ActionScript 编程都涉及创建和处理可视元素，因此存在许多与显示编程有关的任务。本章介绍适用于所有显示对象的常见任务，其中包括：

- 处理显示列表和显示对象容器
 - 在显示列表中添加显示对象
 - 从显示列表中删除对象
 - 在显示容器间移动对象
 - 将对象移到其它对象的前面或后面
- 处理舞台
 - 设置帧速率
 - 控制舞台缩放比例
 - 处理全屏模式
- 处理显示对象事件
- 确定显示对象的位置，包括创建拖放交互组件
- 缩放、旋转显示对象及调整其大小
- 对显示对象应用混合模式、颜色转换和透明度
- 遮罩显示对象
- 对显示对象进行动画处理
- 加载外部显示内容（如 SWF 文件或图像）

本手册中的后面几章将介绍有关处理显示对象的其它任务。这些任务包括适用于任何显示对象的任务和与特定类型的显示对象相关的任务：

- 使用 ActionScript 在显示对象上绘制矢量图形，具体内容请参阅第 265 页的“[使用绘图 API](#)”。
- 对显示对象应用几何转换，具体内容请参阅第 283 页的“[处理几何结构](#)”
- 对显示对象应用图形滤镜效果（如模糊、发光、投影等），具体内容请参阅第 293 页的“[过滤显示对象](#)”
- 使用特定于影片剪辑的特性，具体内容请参阅第 337 页的“[处理影片剪辑](#)”
- 使用 TextField 对象，具体内容请参阅第 359 页的“[处理文本](#)”
- 使用位图图形，具体内容请参阅第 405 页的“[处理位图](#)”
- 使用视频元素，具体内容请参阅第 440 页的“[处理视频](#)”

重要概念和术语

以下参考列表包含将会在本章中遇到的重要术语：

- **Alpha**: 表示颜色中的透明度（更准确地说，是不透明度）的颜色值。例如，Alpha 通道值为 60% 的颜色只显示其最大强度的 60%，即只有 40% 是透明的。
- **位图图形**: 在计算机中定义为彩色像素网格（行和列）的图形。通常，位图图形包括数码照片和类似图像。
- **混合模式**: 指定两个重叠图像的内容应如何进行交互。通常，一个图像上面的另一个不透明图像会遮盖住下面的图像，因此根本看不到该图像；但是，不同的混合模式会导致图像颜色以不同方式混合在一起，因此，生成的内容是两个图像的某种组合形式。

- 显示列表：由 Flash Player 和 AIR 呈现为可见屏幕内容的显示对象的层次结构。舞台是显示列表的根，附加到舞台或其子级之一上的所有显示对象构成了显示列表（即使对象实际上并未呈现，例如，对象位于舞台边界以外）。
- 显示对象：在 Flash Player 或 AIR 中表示某种类型的可视内容的对象。显示列表中只能包含显示对象，所有显示对象类是 `DisplayObject` 类的子类。
- 显示对象容器：一种特殊类型的显示对象，除了（通常）具有其自己的可视表示形式以外，它还可以包含子显示对象。
- SWF 文件的主类：为 SWF 文件中最外面的显示对象定义行为的类，从概念上讲，它是 SWF 文件本身的类。例如，在 Flash 创作工具中创建的 SWF 具有一个“主时间轴”，其中包含所有其它时间轴；SWF 文件的主类是指将主时间轴作为其实例的类。
- 蒙版：一种将图像的某些部分隐藏起来（或者相反，只允许显示图像的某些部分）的技术。遮罩图像部分将变为透明，因此，将显示其下面的内容。此术语与画家所使用的遮蔽胶带非常相似，遮蔽胶带用于防止将颜料喷到某些区域上。
- 舞台：一个可视容器，它是 SWF 文件中的所有可视内容的基础或背景。
- 变形：对图形的可视特性进行的调整，如旋转对象、改变其缩放比例、倾斜或扭曲其形状或者改变其颜色。
- 矢量图形：在计算机中定义为使用特定特性（如粗细、长度、大小、角度以及位置）绘制的线条和形状的图形。

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。由于本章介绍如何创建和操作可视内容，本章中的所有代码清单会实际创建可视对象并将它们显示在屏幕上；测试范例将涉及在 Flash Player 或 AIR 中查看结果，而不是像前几章那样查看变量值。要测试本章中的代码清单，请执行以下操作：


- 1 使用 Flash 创作工具创建空文档
- 2 在时间轴上选择一关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 通过“控制”>“测试影片”运行程序。

您将看到显示在屏幕上的代码结果，并且所有 `trace()` 函数调用将显示在“输出”面板中。

用于测试示例代码清单的方法在第 29 页的“[测试本章内的示例代码清单](#)”中有更详细的介绍。

核心显示类

ActionScript 3.0 的 `flash.display` 包中包括可在 Flash Player 或 AIR 中显示的可视对象的类。下图说明了这些核心显示对象类的子类关系。



该图说明了显示对象类的类继承。请注意，其中某些类，尤其是 `StaticText`、`TextField` 和 `Video` 类，不在 `flash.display` 包中，但它们仍然是从 `DisplayObject` 类继承的。

扩展 `DisplayObject` 类的所有类都继承该类的方法和属性。有关详细信息，请参阅第 231 页的“[DisplayObject 类的属性和方法](#)”。

可以实例化包含在 `flash.display` 包中的下列类的对象：

- `Bitmap` — 使用 `Bitmap` 类可定义从外部文件加载或通过 ActionScript 呈现的位图对象。可以通过 `Loader` 类从外部文件加载位图。可以加载 GIF、JPG 或 PNG 文件。还可以创建包含自定义数据的 `BitmapData` 对象，然后创建使用该数据的 `Bitmap` 对象。可以使用 `BitmapData` 类的方法来更改位图，无论这些位图是加载的还是在 ActionScript 中创建的。有关详细信息，请参阅第 257 页的“[加载显示对象](#)”和第 405 页的“[处理位图](#)”。
- `Loader` — 使用 `Loader` 类可加载外部资源（SWF 文件或图形）。有关详细信息，请参阅第 257 页的“[动态加载显示内容](#)”。
- `Shape` — 使用 `Shape` 类可创建矢量图形，如矩形、直线、圆等。有关详细信息，请参阅第 265 页的“[使用绘图 API](#)”。
- `SimpleButton` — `SimpleButton` 对象是在 Flash 创作工具中创建的按钮元件的 ActionScript 表示形式。`SimpleButton` 实例有四种按钮状态：弹起、按下、指针经过和点击测试（响应鼠标和键盘事件的区域）。
- `Sprite` — `Sprite` 对象可包含它自己的图形，也可包含子显示对象。（`Sprite` 类用于扩展 `DisplayObjectContainer` 类）。有关详细信息，请参阅第 232 页的“[处理显示对象容器](#)”和第 265 页的“[使用绘图 API](#)”。
- `MovieClip` — `MovieClip` 对象是在 Flash 创作工具中创建的影片剪辑元件的 ActionScript 形式。实际上，`MovieClip` 与 `Sprite` 对象类似，不同的是它还有一个时间轴。有关详细信息，请参阅第 337 页的“[处理影片剪辑](#)”。

下列类不在 `flash.display` 包中，这些类是 `DisplayObject` 类的子类：

- `TextField` 类包括在 `flash.text` 包中，它是用于文本显示和输入的显示对象。有关详细信息，请参阅第 359 页的“[处理文本](#)”。
- `Video` 类包括在 `flash.media` 包中，它是用于显示视频文件的显示对象。有关详细信息，请参阅第 440 页的“[处理视频](#)”。

`flash.display` 包中的下列类用于扩展 `DisplayObject` 类，但您不能创建这些类的实例。这些类而是用作其它显示对象的父类，因此可将通用功能合并到一个类中。

- `AVM1Movie` — `AVM1Movie` 类用于表示在 ActionScript 1.0 和 2.0 中创作的已加载 SWF 文件。
- `DisplayObjectContainer` — `Loader`、`Stage`、`Sprite` 和 `MovieClip` 类均用于扩展 `DisplayObjectContainer` 类。有关详细信息，请参阅第 232 页的“[处理显示对象容器](#)”。
- `InteractiveObject` — `InteractiveObject` 是用于与鼠标和键盘交互的所有对象的基类。`SimpleButton`、`TextField`、`Loader`、`Sprite`、`Stage` 和 `MovieClip` 对象是 `InteractiveObject` 类的所有子类。有关创建鼠标和键盘交互的详细信息，请参阅第 498 页的“[捕获用户输入](#)”。
- `MorphShape` — 这些对象是在 Flash 创作工具中创建补间形状时创建的。无法使用 ActionScript 实例化这些对象，但可以从显示列表中访问它们。
- `Stage` — `Stage` 类用于扩展 `DisplayObjectContainer` 类。有一个应用程序的 `Stage` 实例，该实例位于显示列表层次结构的顶部。要访问 `Stage`，请使用任何 `DisplayObject` 实例的 `stage` 属性。有关详细信息，请参阅第 236 页的“[设置舞台属性](#)”。

此外，`flash.text` 包中的 `StaticText` 类也用于扩展 `DisplayObject` 类，但不能在代码中创建它的实例。只能在 Flash 中创建静态文本字段。

显示列表方法的优点

在 ActionScript 3.0 中，不同类型的显示对象有不同的类。在 ActionScript 1.0 和 2.0 中，很多相同类型的对象都包括在一个类（即 `MovieClip` 类）中。

类的这种个性化和显示列表的层次结构有下列优点：

- 呈现方式更为有效且减少了内存使用
- 改进了深度管理
- 完整遍历显示列表
- 列表外的显示对象
- 更便于创建显示对象的子类

呈现方式更为有效且文件较小

在 ActionScript 1.0 和 2.0 中，只可以在 `MovieClip` 对象中绘制形状。在 ActionScript 3.0 中，提供了可在其中绘制形状的更简单的显示对象类。由于这些 ActionScript 3.0 显示对象类并不包括 `MovieClip` 对象中包含的全部方法和属性，因此给内存和处理器资源造成的负担比较小。

例如，每个 `MovieClip` 对象都包括用于影片剪辑时间轴的属性，而 `Shape` 对象则不包括。用于管理时间轴的属性会使用大量的内存和处理器资源。在 ActionScript 3.0 中，使用 `Shape` 对象可提高性能。与更复杂的 `MovieClip` 对象相比，`Shape` 对象的开销更少。Flash Player 和 AIR 并不需要管理未使用的 `MovieClip` 属性，因此提高了速度，还减少了对象使用的内存空间。

改进了深度管理

在 ActionScript 1.0 和 2.0 中，通过线性深度管理方案和方法（如 `getNextHighestDepth()`）进行深度管理。

在 ActionScript 3.0 中提供了 `DisplayObjectContainer` 类，该类提供用于管理显示对象深度的更便捷的方法和属性。

在 ActionScript 3.0 中，当您将显示对象移到 `DisplayObjectContainer` 实例的子级列表中的新位置时，显示对象容器中的其它子级会自动重新定位并在显示对象容器中分配相应的子索引位置。

此外，在 ActionScript 3.0 中，总是可以发现任何显示对象容器中的所有子对象。每个 `DisplayObjectContainer` 实例都有 `numChildren` 属性，用于列出显示对象容器中的子级数。由于显示对象容器的子级列表始终是索引列表，因此可以检查列表中从索引位置 0 到最后一个索引位置 (`numChildren - 1`) 的所有对象。这不适用于 ActionScript 1.0 和 2.0 中 `MovieClip` 对象的方法和属性。

在 ActionScript 3.0 中，可以按顺序轻松遍历显示列表；显示对象容器子级列表的索引编号没有中断。遍历显示列表和管理对象深度比在 ActionScript 1.0 和 2.0 中更容易。在 ActionScript 1.0 和 2.0 中，影片剪辑可以包含深度顺序有间歇中断的对象，这可能导致难以遍历对象列表。在 ActionScript 3.0 中，显示对象容器的每个子级列表都在内部缓存为一个数组，这样按索引查找的速度就非常快。遍历显示对象容器所有子级的速度也非常快。

在 ActionScript 3.0 中，还可以通过使用 `DisplayObjectContainer` 类的 `getChildByName()` 方法来访问显示对象容器中的子级。

完整遍历显示列表

在 ActionScript 1.0 和 2.0 中，无法访问在 Flash 创作工具中绘制的某些对象（如矢量形状）。在 ActionScript 3.0 中，可以访问显示列表中的所有对象，包括使用 ActionScript 创建的对象以及在 Flash 创作工具中创建的所有显示对象。有关详细信息，请参阅第 235 页的“[遍历显示列表](#)”。

列表外的显示对象

在 ActionScript 3.0 中，可以创建不在可视显示列表中的显示对象。这些对象称为“列表外”显示对象。仅当调用已添加到显示列表中的 `DisplayObjectContainer` 实例的 `addChild()` 或 `addChildAt()` 方法时，才会将显示对象添加到可视显示列表中。

可以使用列表外的显示对象来组合复杂的显示对象，如有多个显示对象容器（包含多个显示对象）的那些对象。通过将显示对象放在列表外，可以组合复杂的对象，而不需要占用处理时间来呈现这些显示对象。然后在需要时可以在显示列表中添加列表外的显示对象。此外，可以随意将显示对象容器的子级移入和移出显示列表以及移到显示列表中的任何需要位置。

更便于创建显示对象的子类

在 ActionScript 1.0 和 2.0 中，通常必须在 SWF 文件中添加新的 `MovieClip` 对象才能创建基本形状或显示位图。在 ActionScript 3.0 中，`DisplayObject` 类包括许多内置子类，包括 `Shape` 和 `Bitmap`。由于 ActionScript 3.0 中的类更专用于特定类型的对象，因此更易于创建内置类的基本子类。

例如，要在 ActionScript 2.0 中绘制一个圆，可以在实例化自定义类的对象时创建用于扩展 `MovieClip` 类的 `CustomCircle` 类。但是，该类还另外包括 `MovieClip` 类中不应用于该类的许多属性和方法（如 `totalFrames`）。但在 ActionScript 3.0 中，可以创建用于扩展 `Shape` 对象的 `CustomCircle` 类，但该类不包括 `MovieClip` 类中包含的不相关的属性和方法。下面的代码显示了 `CustomCircle` 类的一个示例：

```

import flash.display.*;

public class CustomCircle extends Shape
{
    var xPos:Number;
    var yPos:Number;
    var radius:Number;
    var color:uint;
    public function CustomCircle(xInput:Number,
                                  yInput:Number,
                                  rInput:Number,
                                  colorInput:uint)
    {
        xPos = xInput;
        yPos = yInput;
        radius = rInput;
        color = colorInput;
        this.graphics.beginFill(color);
        this.graphics.drawCircle(xPos, yPos, radius);
    }
}

```

处理显示对象

现在您已了解了舞台、显示对象、显示对象容器和显示列表的基本概念，本部分将为您提供有关在 ActionScript 3.0 中处理显示对象的一些更具体的信息。

DisplayObject 类的属性和方法

所有显示对象都是 `DisplayObject` 类的子类，同样它们还会继承 `DisplayObject` 类的属性和方法。继承的属性是适用于所有显示对象的基本属性。例如，每个显示对象都有 `x` 属性和 `y` 属性，用于指定对象在显示对象容器中的位置。

您不能使用 `DisplayObject` 类构造函数来创建 `DisplayObject` 实例。必须创建另一种对象（属于 `DisplayObject` 类的子类的对象，如 `Sprite`）才能使用 `new` 运算符来实例化对象。此外，如果要创建自定义显示对象类，还必须创建具有可用构造函数的其中一个显示对象子类（如 `Shape` 类或 `Sprite` 类）。有关详细信息，请参阅《ActionScript 3.0 语言和组件参考》中的 `DisplayObject` 类说明。

在显示列表中添加显示对象

实例化显示对象时，在将显示对象实例添加到显示列表上的显示对象容器之前，显示对象不会出现在屏幕上（即在舞台上）。例如，在下面的代码中，如果省略了最后一行代码，则 `myText` `TextField` 对象不可见。在最后一行代码中，`this` 关键字必须引用已添加到显示列表中的显示对象容器。

```

import flash.display.*;
import flash.text.TextField;
var myText:TextField = new TextField();
myText.text = "Buenos dias.";
this.addChild(myText);

```

当在舞台上添加任何可视元素时，该元素会成为 `Stage` 对象的“子级”。应用程序中加载的第一个 SWF 文件（例如，HTML 页中嵌入的文件）会自动添加为 `Stage` 的子级。它可以是扩展 `Sprite` 类的任何类型的对象。


不是使用 ActionScript（例如，通过在 Adobe Flex Builder 3 中添加 MXML 标签或在 Flash 的舞台上放置项目）创建的任何显示对象都会添加到显示列表中。尽管没有通过 ActionScript 添加这些显示对象，但仍可通过 ActionScript 访问它们。例如，下面的代码将调整在创作工具中（不是通过 ActionScript）添加的名为 `button1` 的对象的宽度：

```
button1.width = 200;
```

处理显示对象容器

如果从显示列表中删除某个 `DisplayObjectContainer` 对象，或者以其它某种方式移动该对象或对其进行变形处理，则会同时删除、移动 `DisplayObjectContainer` 中的每个显示对象或对其进行变形处理。

显示对象容器本身就是一种显示对象，它可以添加到其它显示对象容器中。例如，下图显示的是显示对象容器 `pictureScreen`，它包含一个轮廓形状和四个其它显示对象容器（类型为 `PictureFrame`）：



A. 定义 `pictureScreen` 显示对象容器边界的形状 B. 作为 `pictureScreen` 对象的子级的四个显示对象容器

要使某一显示对象出现在显示列表中，必须将该显示对象添加到显示列表上的显示对象容器中。使用容器对象的 `addChild()` 方法或 `addChildAt()` 方法可执行此操作。例如，如果下面的代码没有最后一行，将不会显示 `myTextField` 对象：

```
var myTextField:TextField = new TextField();
myTextField.text = "hello";
this.root.addChild(myTextField);
```

在此代码范例中，`this.root` 指向包含该代码的 `MovieClip` 显示对象容器。在实际代码中，可以指定其它容器。

使用 `addChildAt()` 方法可将子级添加到显示对象容器的子级列表中的特定位置。子级列表中这些从 0 开始的索引位置与显示对象的分层（从前到后顺序）有关。例如，请考虑下列三个显示对象。每个对象都是从称为 Ball 的自定义类创建的。




使用 `addChildAt()` 方法可以调整这些显示对象在容器中的分层。例如，请考虑使用以下代码：

```
ball_A = new Ball(0xFFCC00, "a");
ball_A.name = "ball_A";
ball_A.x = 20;
ball_A.y = 20;
container.addChild(ball_A);

ball_B = new Ball(0xFFCC00, "b");
ball_B.name = "ball_B";
ball_B.x = 70;
ball_B.y = 20;
container.addChild(ball_B);

ball_C = new Ball(0xFFCC00, "c");
ball_C.name = "ball_C";
ball_C.x = 40;
ball_C.y = 60;
container.addChildAt(ball_C, 1);
```

执行此代码后，显示对象在 `container DisplayObjectContainer` 对象中的定位如下所示。请注意对象的分层。



要重新将对象定位到显示列表的顶部，只需重新将其添加到列表中。例如，在前面的代码后，要将 `ball_A` 移到堆栈的顶部，请使用下面的代码行：

```
container.addChild(ball_A);
```

此代码可有效地将 `ball_A` 从它在 `container` 的显示列表中的位置删除，然后将它重新添加到列表的顶部，最终的结果是将它移到堆栈的顶部。

可以使用 `getChildAt()` 方法来验证显示对象的图层顺序。`getChildAt()` 方法根据您向容器传递的索引编号返回容器的子对象。例如，下面的代码显示 `container DisplayObjectContainer` 对象的子级列表中不同位置的显示对象的名称：

```
trace(container.getChildAt(0).name); // ball_A
trace(container.getChildAt(1).name); // ball_C
trace(container.getChildAt(2).name); // ball_B
```

如果从父容器的子级列表中删除了一个显示对象，则列表中更高位置的每一个元素在子索引中会分别下移一个位置。例如，接着前面的代码，下面的代码显示如果删除子级列表中位置较低的一个显示对象，`container` `DisplayObjectContainer` 中位置 2 的显示对象如何移到位置 1：

```
container.removeChild(ball_C);
trace(container.getChildAt(0).name); // ball_A
trace(container.getChildAt(1).name); // ball_B
```

`removeChild()` 和 `removeChildAt()` 方法并不完全删除显示对象实例。这两种方法只是从容器的子级列表中删除显示对象实例。该实例仍可由另一个变量引用。（请使用 `delete` 运算符完全删除对象。）

由于显示对象只有一个父容器，因此只能在一个显示对象容器中添加显示对象的实例。例如，下面的代码说明了显示对象 `tf1` 只能存在于一个容器中（本例中为 `Sprite`，它扩展 `DisplayObjectContainer` 类）：

```
tf1:TextField = new TextField();
tf2:TextField = new TextField();
tf1.name = "text 1";
tf2.name = "text 2";

container1:Sprite = new Sprite();
container2:Sprite = new Sprite();

container1.addChild(tf1);
container1.addChild(tf2);
container2.addChild(tf1);

trace(container1.numChildren); // 1
trace(container1.getChildAt(0).name); // text 2
trace(container2.numChildren); // 1
trace(container2.getChildAt(0).name); // text 1
```

如果将包含在一个显示对象容器中的显示对象添加到另一个显示对象容器中，则会从第一个显示对象容器的子级列表中删除该显示对象。

除了上面介绍的方法之外，`DisplayObjectContainer` 类还定义了用于处理子显示对象的几个方法，其中包括：


- `contains()`: 确定显示对象是否是 `DisplayObjectContainer` 的子级。
- `getChildByName()`: 按名称检索显示对象。
- `getChildIndex()`: 返回显示对象的索引位置。
- `setChildIndex()`: 更改子显示对象的位置。
- `swapChildren()`: 交换两个显示对象的前后顺序。
- `swapChildrenAt()`: 交换两个显示对象的前后顺序（由其索引值指定）。

有关详细信息，请参阅 [ActionScript 3.0 语言和组件参考](#) 中的相关条目。

回想一下，不在显示列表中的显示对象（即不包括在作为舞台子级的显示对象容器中的显示对象）称为“列表外”显示对象。

遍历显示列表

如您所见，显示列表是一个树结构。树的顶部是舞台，它可以包含多个显示对象。那些本身就是显示对象容器的显示对象可以包含其它显示对象或显示对象容器。



`DisplayObjectContainer` 类包括通过显示对象容器的子级列表遍历显示列表的属性和方法。例如，考虑下面的代码，其中在 `container` 对象（该对象为 `Sprite`，`Sprite` 类用于扩展 `DisplayObjectContainer` 类）中添加了两个显示对象 `title` 和 `pict`：

```

var container:Sprite = new Sprite();
var title:TextField = new TextField();
title.text = "Hello";
var pict:Loader = new Loader();
var url:URLRequest = new URLRequest("banana.jpg");
pict.load(url);
pict.name = "banana loader";
container.addChild(title);
container.addChild(pict);
  
```

`getChildAt()` 方法返回显示列表中特定索引位置的子级：

```
trace(container.getChildAt(0) is TextField); // true
```

您也可以按名称访问子对象。每个显示对象都有一个名称属性；如果没有指定该属性，Flash Player 或 AIR 会指定一个默认值，如 `"instance1"`。例如，下面的代码说明了如何使用 `getChildByName()` 方法来访问名为 `"banana loader"` 的子显示对象：

```
trace(container.getChildByName("banana loader") is Loader); // true
```

与使用 `getChildAt()` 方法相比，使用 `getChildByName()` 方法会导致性能降低。

由于显示对象容器可以包含其它显示对象容器作为其显示列表中的子对象，因此您可将应用程序的完整显示列表作为树来遍历。例如，在前面说明的代码摘录中，完成 `pict` `Loader` 对象的加载操作后，`pict` 对象将加载一个子显示对象，即位图。要访问此位图显示对象，可以编写 `pict.getChildAt(0)`。还可以编写 `container.getChildAt(0).getChildAt(0)`（由于 `container.getChildAt(0) == pict`）。

下面的函数提供了显示对象容器中显示列表的缩进式 `trace()` 输出：

```
function traceDisplayList(container:DisplayObjectContainer, indentString:String = ""):void
{
    var child:DisplayObject;
    for (var i:uint=0; i < container.numChildren; i++)
    {
        child = container.getChildAt(i);
        trace(indentString, child, child.name);
        if (container.getChildAt(i) is DisplayObjectContainer)
        {
            traceDisplayList(DisplayObjectContainer(child), indentString + "")
        }
    }
}
```

设置舞台属性

`Stage` 类用于覆盖 `DisplayObject` 类的大多数属性和方法。如果调用其中一个被覆盖的属性或方法，Flash Player 和 AIR 会引发异常。例如，`Stage` 对象不具有 `x` 或 `y` 属性，因为作为应用程序的主容器，该对象的位置是固定的。`x` 和 `y` 属性是指显示对象相对于其容器的位置，因为舞台没有包含在其它显示对象容器中，所以这些属性不适用。

注：`Stage` 类的某些属性和方法只适用于与加载的第一个 SWF 文件在同一个安全沙箱中的显示对象。有关详细信息，请参阅第 601 页的“[Stage 安全性](#)”。

控制回放帧速率

`Stage` 类的 `framerate` 属性用于设置加载到应用程序中的所有 SWF 文件的帧速率。有关详细信息，请参阅 [ActionScript 3.0 语言和组件参考](#)。

控制舞台缩放比例

当调整呈现 Flash Player 或 AIR 的屏幕部分的大小时，Flash Player 或 AIR 会自动调整舞台内容来加以补偿。`Stage` 类的 `scaleMode` 属性可确定如何调整舞台内容。此属性可以设置为四个不同值，如 `flash.display.StageScaleMode` 类中的常量所定义。

对于三个 `scaleMode` 值 (`StageScaleMode.EXACT_FIT`、`StageScaleMode.SHOW_ALL` 和 `StageScaleMode.NO_BORDER`)，Flash Player 和 AIR 会缩放舞台的内容以使其容纳在舞台的边界内。这三个选项的不同之处在于确定执行缩放的方式：

- `StageScaleMode.EXACT_FIT` 按比例缩放 SWF。
- `StageScaleMode.SHOW_ALL` 确定是否显示边框（就像在标准电视上观看宽屏电影时显示的黑条）。
- `StageScaleMode.NO_BORDER` 确定是否可以部分裁切内容。

或者，如果将 `scaleMode` 设置为 `StageScaleMode.NO_SCALE`，则当查看者调整 Flash Player 或 AIR 窗口大小时，舞台内容将保持定义的大小。仅在此缩放模式中，`Stage` 类的 `stageWidth` 和 `stageHeight` 属性才能用于确定窗口调整大小后的实际像素尺寸。（在其它缩放模式中，`stageWidth` 和 `stageHeight` 属性始终反映的是 SWF 的原始宽度和高度。）此外，当 `scaleMode` 设置为 `StageScaleMode.NO_SCALE` 并且调整了 SWF 文件大小时，将调度 `Stage` 类的 `resize` 事件，以允许您进行相应地调整。

因此，将 `scaleMode` 设置为 `StageScaleMode.NO_SCALE` 可以更好地控制如何根据需要调整屏幕内容以适合窗口大小。例如，在包含视频和控制栏的 SWF 中，您可能希望在调整舞台大小时控制栏的大小保持不变，而仅更改视频窗口大小以适应舞台大小的更改。以下示例中演示了这一点：

```
// videoScreen is a display object (e.g. a Video instance) containing a
// video; it is positioned at the top-left corner of the Stage, and
// it should resize when the SWF resizes.

// controlBar is a display object (e.g. a Sprite) containing several
// buttons; it should stay positioned at the bottom-left corner of the
// Stage (below videoScreen) and it should not resize when the SWF
// resizes.

import flash.display.Stage;
import flash.display.StageAlign;
import flash.display.StageScaleMode;
import flash.events.Event;

var swfStage:Stage = videoScreen.stage;
swfStage.scaleMode = StageScaleMode.NO_SCALE;
swfStage.align = StageAlign.TOP_LEFT;

function resizeDisplay(event:Event):void
{
    var swfWidth:int = swfStage.stageWidth;
    var swfHeight:int = swfStage.stageHeight;

    // Resize the video window.
    var newVideoHeight:Number = swfHeight - controlBar.height;
    videoScreen.height = newVideoHeight;
    videoScreen.scaleX = videoScreen.scaleY;

    // Reposition the control bar.
    controlBar.y = newVideoHeight;
}

swfStage.addEventListener(Event.RESIZE, resizeDisplay);
```

处理全屏模式

使用全屏模式可以将影片的舞台设置为填充查看者的整个显示器，而不包含任何边框或菜单。Stage 类的 displayState 属性用于切换 SWF 的全屏模式。可以将 displayState 属性设置为由 flash.display.StageDisplayState 类中的常量定义的其中一个值。若要打开全屏模式，请将 displayState 属性设置为 StageDisplayState.FULL_SCREEN：

```
stage.displayState = StageDisplayState.FULL_SCREEN;
```

在 Flash Player 中，只能通过 ActionScript 响应鼠标单击（包括右键单击）或按键来启动全屏模式。在应用程序安全沙箱中运行的 AIR 内容不要求在响应用户操作时进入全屏模式。

若要退出全屏模式，请将 displayState 属性设置为 StageDisplayState.NORMAL。

```
stage.displayState = StageDisplayState.NORMAL;
```

此外，用户可以通过将焦点切换到其它窗口或使用以下组合键之一退出全屏模式：Esc 键（所有平台）、Ctrl-W（Windows）、Command-W（Mac）或 Alt-F4（Windows）。

启用 Flash Player 中的全屏模式

若要为 HTML 页中嵌入的 SWF 文件启用全屏模式，必须在嵌入 Flash Player 的 HTML 代码中加入含有名称 allowFullScreen 和值 true 的 param 标签和 embed 属性，如下所示：

```
<object>
    ...
    <param name="allowFullScreen" value="true" />
    <embed ... allowfullscreen="true" />
</object>
```

在 Flash 创作工具中选择“文件”->“发布设置”，并在“发布设置”对话框中的“HTML”选项卡上选择“仅 Flash - 允许全屏”模板。

在 Flex 中，确保 HTML 模板包含支持全屏的 <object> 和 <embed> 标签。

如果要在网页中使用 JavaScript 来生成 SWF 嵌入标签，则必须更改 JavaScript 以添加 allowFullScreen param 标签和属性。例如，如果 HTML 页使用 AC_FL_RunContent() 函数（由 Flex Builder 和 Flash 生成的 HTML 页使用），则应在该函数调用中添加 allowFullScreen 参数，如下所示：

```
AC_FL_RunContent(
    ...
    'allowFullScreen', 'true',
    ...
); //end AC code
```

这不适用于在独立 Flash Player 中运行的 SWF 文件。

注：如果将窗口模式（HTML 中的 wmode）设置为不透明无窗口（opaque）或透明无窗口（transparent），则全屏窗口始终是不透明的。

对浏览器中的 Flash Player 使用全屏模式时还有一些安全方面的限制。第 587 页的“[Flash Player 安全性](#)”中对这些限制进行了说明。

全屏舞台大小和缩放

Stage.fullScreenHeight 和 Stage.fullScreenWidth 属性返回在转为全屏大小时所使用的显示器的高度和宽度（如果是立即进入全屏状态）。在您检索到这些值之后，但在进入全屏模式之前，如果用户有机会将浏览器从一台显示器移至另一台显示器，则这些值可能不正确。如果是在将 Stage.displayState 属性设置为 StageDisplayState.FULL_SCREEN 的同一个事件处理函数中检索这些值，则这些值正确。对于拥有多台显示器的用户，SWF 内容扩大时只会填充一台显示器。Flash Player 和 AIR 使用度量信息来确定哪个显示器包含 SWF 的最大部分内容，然后使用该显示器提供全屏模式。fullScreenHeight 和 fullScreenWidth 属性反映的只是对于全屏模式所使用的显示器大小。有关详细信息，请参阅《ActionScript 3.0 语言和组件参考》中的 [Stage.fullScreenHeight](#) 和 [Stage.fullScreenWidth](#)。

全屏模式的舞台缩放行为与正常模式下的相同；缩放比例由 Stage 类的 scaleMode 属性控制。如果 scaleMode 属性设置为 StageScaleMode.NO_SCALE，则舞台的 stageWidth 和 stageHeight 属性会发生更改，以反映 SWF 所占用的屏幕区域的大小（在本例中为整个屏幕）；如果在浏览器中查看，则此属性的 HTML 参数用于控制该设置。

打开或关闭全屏模式时，可以使用 Stage 类的 fullScreen 事件来进行检测和响应。例如，进入或退出全屏模式时，您可能需要重新定位、添加或删除屏幕中的项目，如本例中所示：

```
import flash.events.FullScreenEvent;

function fullScreenRedraw(event:FullScreenEvent):void
{
    if (event.fullScreen)
    {
        // Remove input text fields.
        // Add a button that closes full-screen mode.
    }
    else
    {
        // Re-add input text fields.
        // Remove the button that closes full-screen mode.
    }
}

mySprite.stage.addEventListener(FullScreenEvent.FULL_SCREEN, fullScreenRedraw);
```

正如此代码所示，fullScreen 事件的事件对象是 flash.events.FullScreenEvent 类的实例，它包含指示是启用（true）还是禁用（false）全屏模式的 fullScreen 属性。

全屏模式下的键盘支持

当 Flash Player 在浏览器中运行时，全屏模式下将禁用所有与键盘相关的 ActionScript，如 TextField 实例中的键盘事件和文本输入。但有一些例外（启用的键）：

- 经过挑选的非打印键，尤其是方向键、空格键和 Tab 键
- 用于终止全屏模式的快捷键：Esc（Windows 和 Mac）、Ctrl-W（Windows）、Command-W（Mac）和 Alt-F4

对于在独立 Flash Player 或 AIR 中运行的 SWF 内容，不存在这些限制。AIR 支持允许键盘输入的交互式全屏模式。

全屏模式下的硬件缩放

使用 Stage 类的 fullScreenSourceRect 属性可以设置 Flash Player 或 AIR 以将舞台的特定区域放大为全屏幕模式。Flash Player 和 AIR 使用用户计算机上的图形和视频卡进行硬件缩放（如果可用），一般来说显示内容的速度要快于软件缩放。

若要利用硬件缩放功能，请将整个舞台或部分舞台设置为全屏模式。以下 ActionScript 3.0 代码将整个舞台设置为全屏模式：

```
import flash.geom.*;
{
    stage.fullScreenSourceRect = new Rectangle(0,0,320,240);
    stage.displayState = StageDisplayState.FULL_SCREEN;
}
```

如果将此属性设置为有效矩形并将 displayState 属性设置为全屏模式，Flash Player 和 AIR 对指定区域进行缩放。

ActionScript 中的实际舞台大小（以像素为单位）不会发生改变。Flash Player 和 AIR 对矩形大小强制规定了最小限制，以容纳标准的“按 Esc 退出全屏模式”消息。通常，此限制大约为 260 x 30 个像素，但可能因平台和 Flash Player 版本而异。

 仅当 Flash Player 或 AIR 处于非全屏模式时，才能设置 fullScreenSourceRect 属性。若要正确使用此属性，请先设置此属性，然后再将 displayState 属性设置为全屏模式。

若要启用缩放功能，请将 fullScreenSourceRect 属性设置为矩形对象。

```
stage.fullScreenSourceRect = new Rectangle(0,0,320,240);
```

若要禁用缩放功能，请将 fullScreenSourceRect 属性设置为 null。

```
stage.fullScreenSourceRect = null;
```

若要利用 Flash Player 中的所有硬件加速功能，请通过 Flash Player 的“设置”对话框将其启用。若要加载该对话框，请在浏览器的 Flash Player 内容中单击右键（Windows）或按住 Control 的同时单击（Mac）。选择“显示”选项卡（第一个选项卡），然后选中复选框：“启用硬件加速”。

直接窗口模式和 GPU 合成窗口模式

Flash Player 10 引入了两种窗口模式：直接模式和 GPU 合成模式。可以通过 Flash 创作工具中的发布设置启用这些模式。

AIR 中不支持这两种模式。若要利用这两种模式，必须为 Flash Player 启用硬件加速。

直接模式使用最快、最直接的路径将图形推送至屏幕，这有利于视频播放。

GPU 合成模式使用显卡中的图形处理单元来加速合成。视频合成是层叠多幅图像以创建单幅视频图像的过程。当采用 GPU 加速合成时，这可提高 YUV 转换、颜色校正、旋转或缩放以及混合的性能。YUV 转换是指将用于传输的合成模拟信号转换为视频摄像头和显示器所使用的 RGB（红、绿、蓝）颜色模型的颜色转换。使用 GPU 加速合成可减少内存占用量以及 CPU 的计算负担。这还促成了标准清晰度视频更平滑的播放。

在实施这些窗口模式时应谨慎小心。使用 GPU 合成可能消耗很多内存资源和 CPU 资源。如果某些操作（如混合模式、过滤、剪裁或遮罩）无法在 GPU 中执行，则可以通过软件来完成。Adobe 建议在使用这些模式时，应限制为每个 HTML 页面一个 SWF 文件，并且不应对横幅启用这些模式。Flash“测试影片”工具不使用硬件加速，但是可通过“发布预览”选项来使用硬件加速。

将 SWF 文件中的帧速率设置为高于 60（最大屏幕刷新率）将不起作用。将帧速率设置为 50 到 55 之间将允许放弃的帧，这种情况经常由于各种原因而发生。

使用直接模式对于 Windows 需要 Microsoft DirectX 9 以及 128 MB VRAM，对于 Apple Macintosh Mac OS X 10.2 版或更高版本需要 OpenGL。GPU 合成模式需要在具有 128 MB VRAM 的 Windows 上支持 Microsoft DirectX 9 和 Pixel Shader 2.0。在 Mac OS X 和 Linux 上，GPU 合成模式需要 OpenGL 1.5 以及多种 OpenGL 扩展（帧缓冲区对象、多纹理、着色器对象、着色语言和片断着色器）。


可通过 Flash“发布设置”对话框，使用“Flash”选项卡上的“硬件加速”菜单，为每个 SWF 单独激活 direct 和 gpu 加速模式。如果选择“无”，则窗口模式将按照“HTML”选项卡上“窗口模式”设置所指定的项，转换为 default、transparent 或 opaque。

处理显示对象的事件

DisplayObject 类从 EventDispatcher 类继承。这意味着，每个显示对象都可完全参与到事件模型中（在第 207 页的“[处理事件](#)”中介绍）。每个显示对象都可使用其 addEventListener() 方法（继承自 EventDispatcher 类）来侦听特定的事件，但仅当侦听对象是该事件的事件流的一部分时才能实现此功能。

当 Flash Player 或 AIR 调度某个事件对象时，该事件对象会执行从舞台到发生事件的显示对象的往返行程。例如，如果用户单击名为子级 1 的显示对象，Flash Player 会从舞台通过显示列表层次结构将事件对象向下调度到子级 1 显示对象。

从概念上说，事件流分为三个阶段，如下图所示：



有关详细信息，请参阅第 207 页的“[处理事件](#)”。

使用显示对象事件时需要记住的一个重要问题是：从显示列表中删除显示对象时，事件侦听器的存在将会对是否从内存中自动删除显示对象（垃圾回收）产生影响。如果显示对象拥有订阅为其事件的侦听器的对象，即使从显示列表中删除了显示对象，也不会从内存中删除显示对象，因为显示对象仍然拥有对这些侦听器对象的引用。有关详细信息，请参阅第 217 页的“[管理事件侦听器](#)”。

选择 DisplayObject 子类

可选的子类有多个，使用显示对象时要做出的一个重要决策是：每个显示对象的用途是什么。以下原则可以帮助您作出决策。无论是需要类实例，还是选择要创建的类的基类，这些建议都适用：

- 如果不需要可作为其它显示对象的容器的对象（即只需要用作独立屏幕元素的对象），请根据用途选择 DisplayObject 或 InteractiveObject 两个子类中的一个：
 - 用于显示位图图像的 Bitmap。
 - 用于添加文本的 TextField。
 - 用于显示视频的 Video。
 - 用于绘制屏幕内容的“画布”的 Shape。特别是，如果要创建用于在屏幕上绘制形状的实例，而该实例不是其它显示对象的容器，则使用 Shape 比使用 Sprite 或 MovieClip 有明显的性能优势。

- 用于由 Flash 创作工具创建的项的 MorphShape、StaticText 或 SimpleButton。(无法以编程方式创建这些类的实例，但可以通过创建具有这些数据类型的变量来引用使用 Flash 创作工具创建的项。)
- 如果需要使用变量来引用主舞台，请使用 Stage 类作为其数据类型。
- 如果需要容器来加载外部 SWF 文件或图像文件，请使用 Loader 实例。加载的内容将作为 Loader 实例的子级添加到显示列表中。其数据类型将取决于加载内容的性质，如下所示：
 - 加载的图像将是 Bitmap 实例。
 - 使用 ActionScript 3.0 编写的已加载 SWF 文件将是 Sprite 或 MovieClip 实例（或这些类的子类的实例，由内容创建者指定）。
 - 使用 ActionScript 1.0 或 ActionScript 2.0 编写的已加载 SWF 文件将是 AVM1Movie 实例。
- 如果需要将一个对象用作其它显示对象的容器（无论是否还要使用 ActionScript 在显示对象上进行绘制），请选择其中一个 DisplayObjectContainer 子类：
 - 如果对象是只使用 ActionScript 创建的，或者如果对象作为只使用 ActionScript 创建和处理的自定义显示对象的基本类，请选择 Sprite。
 - 如果要通过创建变量来引用在 Flash 创作工具中创建的影片剪辑元件，请选择 MovieClip。
- 如果要创建的类与 Flash 库中的影片剪辑元件关联，请选择其中一个 DisplayObjectContainer 子类作为该类的基本类：
 - 如果关联的影片剪辑元件在多个帧上有内容，请选择 MovieClip
 - 如果关联的影片剪辑元件仅在第一帧上有内容，请选择 Sprite

处理显示对象

无论选择使用哪个显示对象，都会有许多的操作，这些操作作为屏幕上显示的一些元素是所有显示对象共有的。例如，可以在屏幕上确定所有显示对象的位置、前后移动显示对象的堆叠顺序、缩放或旋转显示对象等。因为所有显示对象都从它们共有的基类 (DisplayObject) 继承了此功能，所以无论是要操作 TextField 实例、Video 实例、Shape 实例还是其它任何显示对象，此功能的行为都相同。以下部分将详细说明几个常用显示对象操作。

改变位置

对任何显示对象进行的最基本操作是确定显示对象在屏幕上的位置。若要设置显示对象的位置，请更改对象的 x 和 y 属性。

```
myShape.x = 17;  
myShape.y = 212;
```

显示对象定位系统将舞台视为一个笛卡尔坐标系（带有水平 x 轴和垂直 y 轴的常见网格系统）。坐标系的原点（x 和 y 轴相交的 0,0 坐标）位于舞台的左上角。从原点开始，x 轴的值向右为正，向左为负，而 y 轴的值向下为正，向上为负（与典型的图形系统相反）。例如，通过前面的代码行可以将对象 myShape 移到 x 轴坐标 17（原点向右 17 个像素）和 y 轴坐标 212（原点向下 212 个像素）。

默认情况下，当使用 ActionScript 创建显示对象时，x 和 y 属性均设置为 0，从而可将对象放在其父内容的左上角。

改变相对于舞台的位置

x 和 y 属性始终是指显示对象相对于其父显示对象坐标轴的 0,0 坐标的位置，记住这一点很重要。因此，对于包含在 Sprite 实例内的 Shape 实例（如圆），如果将 Shape 对象的 x 和 y 属性设置为 0，则会将圆放在 Sprite 的左上角，该位置不一定是舞台的左上角。若要确定对象相对于全局舞台坐标的位置，可以使用任何显示对象的 globalToLocal() 方法将坐标从全局（舞台）坐标转换为本地（显示对象容器）坐标，如下所示：

```
// Position the shape at the top-left corner of the Stage,  
// regardless of where its parent is located.  
  
// Create a Sprite, positioned at x:200 and y:200.  
var mySprite:Sprite = new Sprite();  
mySprite.x = 200;  
mySprite.y = 200;  
this.addChild(mySprite);  
  
// Draw a dot at the Sprite's 0,0 coordinate, for reference.  
mySprite.graphics.lineStyle(1, 0x000000);  
mySprite.graphics.beginFill(0x000000);  
mySprite.graphics.moveTo(0, 0);  
mySprite.graphics.lineTo(1, 0);  
mySprite.graphics.lineTo(1, 1);  
mySprite.graphics.lineTo(0, 1);  
mySprite.graphics.endFill();  
  
// Create the circle Shape instance.  
var circle:Shape = new Shape();  
mySprite.addChild(circle);  
  
// Draw a circle with radius 50 and center point at x:50, y:50 in the Shape.  
circle.graphics.lineStyle(1, 0x000000);  
circle.graphics.beginFill(0xff0000);  
circle.graphics.drawCircle(50, 50, 50);  
circle.graphics.endFill();  
  
// Move the Shape so its top-left corner is at the Stage's 0, 0 coordinate.  
var stagePoint:Point = new Point(0, 0);  
var targetPoint:Point = mySprite.globalToLocal(stagePoint);  
circle.x = targetPoint.x;  
circle.y = targetPoint.y;
```

同样，也可以使用 `DisplayObject` 类的 `localToGlobal()` 方法将本地坐标转换为舞台坐标。

创建拖放交互组件

移动显示对象的一个常见理由是要创建拖放交互组件，这样当用户单击某个对象时，在松开鼠标按键之前，该对象会随着鼠标的移动而移动。在 ActionScript 中可以采用两种方法创建拖放交互组件。在每种情况下，都会使用两个鼠标事件：按下鼠标按键时，通知对象跟随鼠标光标；松开鼠标按键时，通知对象停止跟随鼠标光标。

第一方法使用 `startDrag()` 方法，它比较简单，但限制较多。按下鼠标按键时，将调用要拖动的显示对象的 `startDrag()` 方法。松开鼠标按键时，将调用 `stopDrag()` 方法。

```
// This code creates a drag-and-drop interaction using the startDrag()  
// technique.  
// square is a DisplayObject (e.g. a MovieClip or Sprite instance).  
  
import flash.events.MouseEvent;  
  
// This function is called when the mouse button is pressed.  
function startDragging(event:MouseEvent):void  
{  
    square.startDrag();  
}  
  
// This function is called when the mouse button is released.  
function stopDragging(event:MouseEvent):void  
{  
    square.stopDrag();  
}  
  
square.addEventListener(MouseEvent.MOUSE_DOWN, startDragging);  
square.addEventListener(MouseEvent.MOUSE_UP, stopDragging);
```

这种方法有一个非常大的限制：使用 `startDrag()` 时，每次只能拖动一个项目。如果正在拖动一个显示对象，然后对另一个显示对象调用了 `startDrag()` 方法，则第一个显示对象会立即停止跟随鼠标。例如，如果 `startDragging()` 函数如下发生了更改，则只拖动 `circle` 对象，而不管 `square.startDrag()` 方法调用：

```
function startDragging(event:MouseEvent):void  
{  
    square.startDrag();  
    circle.startDrag();  
}
```

由于每次只能使用 `startDrag()` 拖动一个对象，因此，可以对任何显示对象调用 `stopDrag()` 方法，这会停止当前正在拖动的任何对象。

如果需要拖动多个显示对象，或者为了避免因多个对象可能使用 `startDrag()` 而发生冲突，最好使用鼠标跟随方法来创建拖动效果。通过这种技术，当按下鼠标按键时，会将函数作为舞台的 `mouseMove` 事件的侦听器来订阅。然后，每次鼠标移动时都会调用此函数，它将使所拖动的对象跳到鼠标所在的 `x,y` 坐标。松开鼠标按键后，取消此函数作为侦听器的订阅，这意味着鼠标移动时不再调用该函数且对象停止跟随鼠标光标。下面是演示说明此技术的一些代码：

```
// This code creates a drag-and-drop interaction using the mouse-following  
// technique.  
// circle is a DisplayObject (e.g. a MovieClip or Sprite instance).  
  
import flash.events.MouseEvent;  
  
var offsetX:Number;  
var offsetY:Number;  
  
// This function is called when the mouse button is pressed.  
function startDragging(event:MouseEvent):void  
{  
    // Record the difference (offset) between where  
    // the cursor was when the mouse button was pressed and the x, y  
    // coordinate of the circle when the mouse button was pressed.  
    offsetX = event.stageX - circle.x;  
    offsetY = event.stageY - circle.y;  
  
    // tell Flash Player to start listening for the mouseMove event  
    stage.addEventListener(MouseEvent.MOUSE_MOVE, dragCircle);  
}  
  
// This function is called when the mouse button is released.
```

```
function stopDragging(event:MouseEvent):void
{
    // Tell Flash Player to stop listening for the mouseMove event.
    stage.removeEventListener(MouseEvent.MOUSE_MOVE, dragCircle);
}

// This function is called every time the mouse moves,
// as long as the mouse button is pressed down.
function dragCircle(event:MouseEvent):void
{
    // Move the circle to the location of the cursor, maintaining
    // the offset between the cursor's location and the
    // location of the dragged object.
    circle.x = event.stageX - offsetX;
    circle.y = event.stageY - offsetY;

    // Instruct Flash Player to refresh the screen after this event.
    event.updateAfterEvent();
}

circle.addEventListener(MouseEvent.MOUSE_DOWN, startDragging);
circle.addEventListener(MouseEvent.MOUSE_UP, stopDragging);
```

除了使显示对象跟随鼠标光标之外，拖放交互组件的共有部分还包括将拖动对象移到显示对象的前面，以便拖动对象好像浮动在所有其它对象之上。例如，假定您有两个对象（圆和正方形），它们都有一个拖放交互组件。如果圆在显示列表中出现在正方形之下，您单击并拖动圆时光标会出现在正方形之上，圆好像在正方形之后滑动，这样中断了拖放视觉效果。您可以使用拖放交互组件避免这一点，以便在单击圆时圆移到显示列表的顶部，使圆会始终出现在其它任何内容的顶部。

下面的代码（摘自前面的示例）用于创建两个显示对象（圆和正方形）的拖放交互组件。只要在任一个显示对象上按下鼠标按键，该显示对象就会移到舞台显示列表的顶部，所以拖动的项目始终出现在顶部。新代码或以前代码中经过更改的代码用粗体显示。

```
// This code creates a drag-and-drop interaction using the mouse-following
// technique.
// circle and square are DisplayObjects (e.g. MovieClip or Sprite
// instances).

import flash.display.DisplayObject;
import flash.events.MouseEvent;

var offsetX:Number;
var offsetY:Number;
var draggedObject:DisplayObject;

// This function is called when the mouse button is pressed.
function startDragging(event:MouseEvent):void
{
    // remember which object is being dragged
    draggedObject = DisplayObject(event.target);

    // Record the difference (offset) between where the cursor was when
    // the mouse button was pressed and the x, y coordinate of the
    // dragged object when the mouse button was pressed.
    offsetX = event.stageX - draggedObject.x;
    offsetY = event.stageY - draggedObject.y;

    // move the selected object to the top of the display list
    stage.addChild(draggedObject);

    // Tell Flash Player to start listening for the mouseMove event.
    stage.addEventListener(MouseEvent.MOUSE_MOVE, dragObject);
}
```

```
// This function is called when the mouse button is released.  
function stopDragging(event:MouseEvent):void  
{  
    // Tell Flash Player to stop listening for the mouseMove event.  
    stage.removeEventListener(MouseEvent.MOUSE_MOVE, dragObject);  
}  
  
// This function is called every time the mouse moves,  
// as long as the mouse button is pressed down.  
function dragObject(event:MouseEvent):void  
{  
    // Move the dragged object to the location of the cursor, maintaining  
    // the offset between the cursor's location and the location  
    // of the dragged object.  
    draggedObject.x = event.stageX - offsetX;  
    draggedObject.y = event.stageY - offsetY;  
  
    // Instruct Flash Player to refresh the screen after this event.  
    event.updateAfterEvent();  
}  
  
circle.addEventListener(MouseEvent.MOUSE_DOWN, startDragging);  
circle.addEventListener(MouseEvent.MOUSE_UP, stopDragging);  
  
square.addEventListener(MouseEvent.MOUSE_DOWN, startDragging);  
square.addEventListener(MouseEvent.MOUSE_UP, stopDragging);
```

若要进一步扩展这种效果，如在几副纸牌（或几组标记）之间移动纸牌（或标记）的游戏中，您可以在“拿出”拖动对象时将拖动对象添加到舞台的显示列表中，然后在“放入”拖动对象时（通过松开鼠标按键）将拖动对象添加到另一个显示列表中（如“那副纸牌”或“那组标记”）。

最后，要增强效果，您可以在单击显示对象时（开始拖动显示对象时）对显示对象应用投影滤镜，然后在松开对象时删除投影。有关在 ActionScript 中使用投影滤镜和其它显示对象滤镜的详细信息，请参阅第 293 页的“[过滤显示对象](#)”。

平移和滚动显示对象

如果显示对象太大，不能在要显示它的区域中完全显示出来，则可以使用 scrollRect 属性定义显示对象的可查看区域。此外，通过更改 scrollRect 属性响应用户输入，可以使内容左右平移或上下滚动。

scrollRect 属性是 Rectangle 类的实例，Rectangle 类包括将矩形区域定义为单个对象所需的有关值。最初定义显示对象的可查看区域时，请创建一个新的 Rectangle 实例并为该实例分配显示对象的 scrollRect 属性。以后进行滚动或平移时，请将 scrollRect 属性读入单独的 Rectangle 变量，然后更改所需的属性（例如，更改 Rectangle 实例的 x 属性进行平移，或更改 y 属性进行滚动）。然后将该 Rectangle 实例重新分配给 scrollRect 属性，将更改的值通知显示对象。

例如，下面的代码定义名为 bigText 的 TextField 对象的可查看区域，该对象因太高而无法容纳在 SWF 文件的边界内。单击名为 up 和 down 的两个按钮时，它们调用的函数通过修改 scrollRect Rectangle 实例的 y 属性而使 TextField 对象的内容向上或向下滚动。

```

import flash.events.MouseEvent;
import flash.geom.Rectangle;

// Define the initial viewable area of the TextField instance.
// left: 0, top: 0, width: TextField's width, height: 350 pixels.
bigText.scrollRect = new Rectangle(0, 0, bigText.width, 350);

// Cache the TextField as a bitmap to improve performance.
bigText.cacheAsBitmap = true;

// called when the "up" button is clicked
function scrollUp(event:MouseEvent):void
{
    // Get access to the current scroll rectangle.
    var rect:Rectangle = bigText.scrollRect;
    // Decrease the y value of the rectangle by 20, effectively
    // shifting the rectangle down by 20 pixels.
    rect.y -= 20;
    // Reassign the rectangle to the TextField to "apply" the change.
    bigText.scrollRect = rect;
}

// called when the "down" button is clicked
function scrollDown(event:MouseEvent):void
{
    // Get access to the current scroll rectangle.
    var rect:Rectangle = bigText.scrollRect;
    // Increase the y value of the rectangle by 20, effectively
    // shifting the rectangle up by 20 pixels.
    rect.y += 20;
    // Reassign the rectangle to the TextField to "apply" the change.
    bigText.scrollRect = rect;
}

up.addEventListener(MouseEvent.CLICK, scrollUp);
down.addEventListener(MouseEvent.CLICK, scrollDown);

```

正如此示例所示，使用显示对象的 scrollRect 属性时，最好指定 Flash Player 或 AIR 应使用 cacheAsBitmap 属性将显示对象的内容缓存为位图。这样，每次滚动显示对象时，Flash Player 和 AIR 就不必重绘显示对象的整个内容，而只需改用缓存的位图即可将所需部分直接呈现到屏幕上。有关详细信息，请参阅第 248 页的“[缓存显示对象](#)”。

处理大小和缩放对象

有两种方式可测量和操作显示对象的大小：使用尺寸属性（width 和 height）或缩放属性（scaleX 和 scaleY）。

每个显示对象都有 width 属性和 height 属性，它们最初设置为对象的大小，以像素为单位。您可以通过读取这些属性的值来确定显示对象的大小。还可以指定新值来更改对象的大小，如下所示：

```

// Resize a display object.
square.width = 420;
square.height = 420;

// Determine the radius of a circle display object.
var radius:Number = circle.width / 2;

```

更改显示对象的 height 或 width 会导致缩放对象，这意味着对象内容将经过伸展或挤压以适合新区域的大小。如果显示对象仅包含矢量形状，将按新缩放比例重绘这些形状，而品质不变。此时将缩放显示对象中的所有位图图形元素，而不是重绘。例如，缩放图形时，如果数码照片的宽度和高度增加后超出图像中像素信息的实际大小，数码照片将被像素化，使数码照片显示带有锯齿。

当更改显示对象的 width 或 height 属性时，Flash Player 和 AIR 也会更新对象的 scaleX 和 scaleY 属性。

注：**TextField** 对象是此缩放行为的例外。文本字段需要调整自身大小，以适应文本自动换行和字体大小，因此文本字段在调整大小之后将其 **scaleX** 或 **scaleY** 值重置为 1。但是，如果调整 **TextField** 对象的 **scaleX** 或 **scaleY** 值，则宽度和高度值会更改，以适应您提供的缩放值。

这些属性表示显示对象与其原始大小相比的相对大小。**scaleX** 和 **scaleY** 属性使用小数（十进制）值来表示百分比。例如，如果某个显示对象的 **width** 已更改，其宽度是原始大小的一半，则该对象的 **scaleX** 属性的值为 .5，表示 50%。如果其高度加倍，则其 **scaleY** 属性的值为 2，表示 200%。

```
// circle is a display object whose width and height are 150 pixels.  
// At original size, scaleX and scaleY are 1 (100%).  
trace(circle.scaleX); // output: 1  
trace(circle.scaleY); // output: 1  
  
// When you change the width and height properties,  
// Flash Player changes the scaleX and scaleY properties accordingly.  
circle.width = 100;  
circle.height = 75;  
trace(circle.scaleX); // output: 0.6622516556291391  
trace(circle.scaleY); // output: 0.4966887417218543
```

此时，大小更改不成比例。换句话说，如果更改一个正方形的 **height** 但不更改其 **width**，则其边长不再相同，它将是一个矩形而不是一个正方形。如果要更改显示对象的相对大小，则可以通过设置 **scaleX** 和 **scaleY** 属性的值来调整该对象的大小，另一种方法是设置 **width** 或 **height** 属性。例如，下面的代码将更改名为 **square** 的显示对象的 **width**，然后更改垂直缩放 (**scaleY**) 以匹配水平缩放，所以正方形的大小成比例。

```
// Change the width directly.  
square.width = 150;  
  
// Change the vertical scale to match the horizontal scale,  
// to keep the size proportional.  
square.scaleY = square.scaleX;
```

控制缩放时的扭曲

通常，缩放显示对象（例如水平伸展）时，引起的扭曲在整个对象上是均匀分布的，所以各部分的伸展量是相同的。对于图形和设计元素，这可能是您所希望的结果。但是，有时更希望能控制显示对象的某些部分伸展、某些部分保持不变。这种情况的一个常见示例是有圆角的矩形按钮。进行正常缩放时，按钮的角将伸展，从而使角半径随按钮大小的调整而改变。



但在这种情况下，最好对缩放进行控制，即能够指定应缩放的某些区域（直边和中间）和不应缩放的区域（角），以便在缩放后不会出现可见的扭曲。



可以使用 9 切片缩放 (Scale-9) 来创建在其中控制如何缩放对象的显示对象。使用 9 切片缩放时，显示对象被分成 9 个单独的矩形（一个 3 x 3 的网格，就像一个“井”字）。矩形的大小不必一定相同，您可以指定放置网格线的位置。缩放显示对象时，四个角矩形中的任何内容（如按钮的圆角）不伸展也不压缩。上中矩形和下中矩形将进行水平缩放，但不进行垂直缩放，而左中矩形和右中矩形将进行垂直缩放，但不进行水平缩放。中心矩形既进行水平缩放又进行垂直缩放。




请记住，如果要创建显示对象且希望某些内容从不缩放，只需要通过放置 9 切片缩放网格的划分线来确保有关内容完全放在其中一个角矩形中即可。


在 ActionScript 中，如果为显示对象的 `scale9Grid` 属性设置一个值，就会打开该对象的 9 切片缩放并定义该对象的缩放 9 网格中矩形的大小。可以使用 `Rectangle` 类的实例作为 `scale9Grid` 属性的值，如下所示：

```
myButton.scale9Grid = new Rectangle(32, 27, 71, 64);
```


`Rectangle` 构造函数的四个参数是 `x` 坐标、`y` 坐标、`width` 和 `height`。在此示例中，矩形的左上角放在名为 `myButton` 的显示对象上的点 `x: 32, y: 27` 处。矩形宽 71 个像素，高 64 个像素（因此其右边位于显示对象上 `x` 坐标为 103 的位置，其下边位于显示对象上 `y` 坐标为 92 的位置）。



包含在由 `Rectangle` 实例定义的区域中的实际区域表示 Scale-9 网格的中心矩形。其它矩形是由 Flash Player 和 AIR 通过扩展 `Rectangle` 实例的各边计算出来的，如下所示：



在本例中，当按钮放大或缩小时，圆角不拉伸也不压缩，但其它区域将通过调整来适应缩放。



A. `myButton.width = 131; myButton.height = 106;` B. `myButton.width = 73; myButton.height = 69;` C. `myButton.width = 54; myButton.height = 141;`

缓存显示对象

如果 Flash 中的设计尺寸增大，无论创建的是应用程序还是复杂的脚本动画，都需要考虑性能和优化。如果内容保持为静态（如矩形 `Shape` 实例），则 Flash Player 和 AIR 不会优化内容。因此，更改矩形的位置时，Flash Player 或 AIR 会重绘整个 `Shape` 实例。

可以通过缓存指定的显示对象来提高 SWF 文件的性能。显示对象是一个“表面”，实际上是位图版本的实例矢量数据，而矢量数据是 SWF 文件中不需要进行太多更改的一种数据。因此，打开缓存的实例不会随 SWF 文件的播放而不断地重绘，这样便可快速呈现 SWF 文件。

注：可以更新矢量数据，这时将重新创建表面。因此，缓存在表面中的矢量数据不需要在整个 SWF 文件中保持一样。

将显示对象的 `cacheAsBitmap` 属性设置为 `true` 会使显示对象缓存其自身的位图表示形式。**Flash Player** 或 AIR 为该实例创建一个表面对象，该对象是一个缓存的位图，而不是矢量数据。如果要更改显示对象的边界，则重新创建表面而不是调整其大小。表面可以嵌套在其它表面之内。子表面会将其位图复制到它的父表面上。有关详细信息，请参阅第 250 页的“[启用位图缓存](#)”。

`DisplayObject` 类的 `opaqueBackground` 属性和 `scrollRect` 属性与使用 `cacheAsBitmap` 属性的位图缓存有关。尽管这三个属性彼此互相独立，但当对象缓存为位图时，`opaqueBackground` 和 `scrollRect` 属性的作用最佳；只有将 `cacheAsBitmap` 设置为 `true` 时，才能看到 `opaqueBackground` 和 `scrollRect` 属性带来的性能优势。有关滚动显示对象内容的详细信息，请参阅第 245 页的“[平移和滚动显示对象](#)”。有关设置不透明背景的详细信息，请参阅第 250 页的“[设置不透明背景颜色](#)”。

有关 Alpha 通道遮罩（要求将 `cacheAsBitmap` 属性设置为 `true`）的信息，请参阅第 254 页的“[遮罩显示对象](#)”。

何时启用缓存

对显示对象启用缓存可创建表面，表面具有助于更快地呈现复杂的矢量动画等优点。有几种情形需要启用缓存。可能您总是希望通过启用缓存来提高 SWF 文件的性能；但是，某些情况下启用缓存并不能提高性能，甚至还会降低性能。本部分介绍在哪些情况下应使用缓存，以及何时使用常规显示对象。

缓存数据的总体性能取决于实例的矢量数据的复杂程度、要更改的数据量，以及是否设置了 `opaqueBackground` 属性。如果要更改的范围较小，则使用表面和使用矢量数据的差异微乎其微。在部署应用程序之前您可能需要实际测试一下这两种情况。

何时使用位图缓存

在以下典型情形中启用位图缓存您可能会看到明显的好处。

- 复杂背景图像：包含矢量数据的详细的复杂背景图像的应用程序（可能是应用了跟踪位图命令的图像，也可能是在 Adobe Illustrator® 中创建的插图）。您可能会在背景上设计动画人物，这会降低动画的速度，因为背景需要持续地重新生成矢量数据。要提高性能，可以将背景显示对象的 `opaqueBackground` 属性设置为 `true`。背景将呈现为位图，可以迅速地重新绘制，以便更快地播放动画。
- 滚动文本字段：应用程序在滚动文本字段中显示大量的文本。可以将文本字段放置在您设置为可滚动的具有滚动框（使用 `scrollRect` 属性）的显示对象中。这可以使指定的实例进行快速像素滚动。当用户滚动显示对象实例时，Flash Player 或 AIR 会通过将滚动的像素向上移来生成新的看得见的区域，而不是重新生成整个文本字段。
- 窗口排列秩序：应用程序具有秩序复杂的重叠窗口。每个窗口都可以打开或关闭（例如，Web 浏览器窗口）。如果将每个窗口标记为一个表面（将 `cacheAsBitmap` 属性设置为 `true`），则各个窗口将隔离开来进行缓存。用户可以拖动窗口使其互相重叠，每个窗口无需重新生成矢量内容。
- Alpha 通道遮罩：当使用 Alpha 通道遮罩时，必须将 `cacheAsBitmap` 属性设置为 `true`。有关详细信息，请参阅第 254 页的“[遮罩显示对象](#)”。

所有这些情况下，启用位图缓存后都通过优化矢量图来提高应用程序的响应能力和互动性。

此外，只要对显示对象应用滤镜，`cacheAsBitmap` 就会自动设置为 `true`，即使将其显式设置为 `false` 也是如此。如果清除了显示对象的所有滤镜，则 `cacheAsBitmap` 属性会返回最后设置的值。

何时避免使用位图缓存

滥用此功能会给 SWF 文件带来负面影响。使用位图缓存时，请记住下面的准则：

- 不要过度使用表面（启用了缓存的显示对象）。每个表面使用的内存都比常规显示对象多，这意味着只在需要提高呈现性能时才启用表面。
- 缓存的位图使用的内存比常规显示对象多很多。例如，如果舞台上 `Sprite` 实例的大小为 250 x 250 个像素，缓存它可能会使用 250 KB 内存；如果它是常规（未缓存的）`Sprite` 实例，则使用 1 KB 内存。
- 避免放大缓存的表面。如果过度使用位图缓存，尤其是在缩小内容时，将使用大量内存（请参阅上一段落）。

- 将表面用于通常为静态（非动画）的显示对象实例。可以拖放或移动实例，但实例的内容不能为动画或更改太多。（动画或变化的内容更可能包含在包含动画的 MovieClip 实例或 Video 实例中。）例如，如果旋转或转换实例，实例将在表面和矢量数据之间进行变化，这种情况难于处理，并会对 SWF 文件产生负面影响。
- 如果将表面与矢量数据混在一起，将增加 Flash Player 和 AIR（有时还包括计算机）的工作量。应尽量将表面归为一组 — 例如，创建窗口应用程序时。

启用位图缓存

要为显示对象启用位图缓存，请将它的 `cacheAsBitmap` 属性设置为 `true`:

```
mySprite.cacheAsBitmap = true;
```

将 `cacheAsBitmap` 属性设置为 `true` 后，您可能会注意到，显示对象的像素会自动与整个坐标对齐。测试 SWF 文件时，您还会注意到，在复杂矢量图像上执行的任何动画的呈现速度都快得多。

即便是将 `cacheAsBitmap` 设置为 `true`，如果出现以下一种或多种情形，也将不创建表面（缓存的位图）：

- 位图高度或宽度超过 2880 像素。
- 位图分配不成功（由于内存不足而出现的错误）。

设置不透明背景颜色

可以为显示对象设置不透明背景。例如，如果 SWF 的背景中包含复杂的矢量图片，则可以将 `opaqueBackground` 属性设置为指定的颜色（通常与舞台颜色相同）。将颜色指定为一个数字（通常为十六进制的颜色值）。然后可将背景视作位图，这样有助于优化性能。

当将 `cacheAsBitmap` 设置为 `true` 并将 `opaqueBackground` 属性设置为指定的颜色时，`opaqueBackground` 属性可以使内部位图不透明而加快呈现速度。如果不将 `cacheAsBitmap` 设置为 `true`，`opaqueBackground` 属性将在显示对象的背景中添加一个不透明的矢量正方形形状。不会自动创建位图。

下面的示例说明了如何设置显示对象的背景以优化性能。

```
myShape.cacheAsBitmap = true;
myShape.opaqueBackground = 0xFF0000;
```

在本例中，将名为 `myShape` 的 `Shape` 的背景颜色设置为红色（`0xFF0000`）。假定 `Shape` 实例在白色背景的舞台上包含一个绿色三角形绘图，这将在 `Shape` 实例的边框（完全包含 `Shape` 的矩形）内显示一个绿色三角形，且空白区域为红色。



当然，如果此代码用于纯红色背景的舞台，则更合理。在其它颜色的背景上，则改为指定该颜色。例如，在白色背景的 SWF 中，`opaqueBackground` 属性最适合设置为 `0xFFFFFFFF` 或纯白色。

应用混合模式

混合模式涉及将一个图像（基图像）的颜色与另一个图像（混合图像）的颜色进行组合来生成第三个图像，所得的图像是实际在屏幕上显示的图像。图像中的每个像素值都会被使用其它图像的对应像素值进行处理，以便在结果的同一位置生成一个像素值。

每个显示对象都有 `blendMode` 属性，可以将其设置为下列混合模式之一。以下是在 `BlendMode` 类中定义的常量。此外，还可以使用 `String` 值（在括号中），这些值是常量的实际值。

- `BlendMode.ADD ("add")`: 通常用于创建两个图像之间的动画变亮模糊效果。

- BlendMode.ALPHA ("alpha"): 通常用于在背景上应用前景的透明度。
- BlendMode.DARKEN ("darker"): 通常用于重叠类型。
- BlendMode.DIFFERENCE ("difference"): 通常用于创建更多变动的颜色。
- BlendMode.ERASE ("erase"): 通常用于使用前景 Alpha 剪掉 (擦除) 背景的一部分。
- BlendMode.HARDLIGHT ("hardlight"): 通常用于创建阴影效果。
- BlendMode.INVERT ("invert"): 用于反转背景。
- BlendMode.LAYER ("layer"): 用于强制为特定显示对象的预构成创建临时缓冲区。
- BlendMode.LIGHTEN ("lighten"): 通常用于重叠类型。
- BlendMode.MULTIPLY ("multiply"): 通常用于创建阴影和深度效果。
- BlendMode.NORMAL ("normal"): 用于指定混合图像的像素值覆盖基本图像的像素值。
- BlendMode.OVERLAY ("overlay"): 通常用于创建阴影效果。
- BlendMode.SCREEN ("screen"): 通常用于创建亮点和镜头眩光。
- BlendMode.SHADER ("shader"): 用于指定用于创建自定义混合效果的 Pixel Bender 着色器。有关使用着色器的详细信息, 请参阅第 319 页的“[使用 Pixel Bender 着色器](#)”。
- BlendMode.SUBTRACT ("subtract"): 通常用于创建两个图像之间的动画变暗模糊效果。

调整 DisplayObject 颜色

可以使用 ColorTransform 类的方法 (`flash.geom.ColorTransform`) 来调整显示对象的颜色。每个显示对象都有 `transform` 属性 (它是 `Transform` 类的实例), 还包含有关应用到显示对象的各种变形的信息 (如旋转、缩放或位置的更改等)。除了有关几何变形的信息之外, `Transform` 类还包括 `colorTransform` 属性, 它是 `ColorTransform` 类的实例, 并提供访问来对显示对象进行颜色调整。要访问显示对象的颜色转换信息, 可以使用如下代码:

```
var colorInfo:ColorTransform = myDisplayObject.transform.colorTransform;
```

创建 `ColorTransform` 实例后, 可以通过读取其属性值来查明已应用了哪些颜色转换, 也可以通过设置这些值来更改显示对象的颜色。要在进行任何更改后更新显示对象, 必须将 `ColorTransform` 实例重新分配给 `transform.colorTransform` 属性。

```
var colorInfo:ColorTransform = myDisplayObject.transform.colorTransform;  
  
// Make some color transformations here.  
  
// Commit the change.  
myDisplayObject.transform.colorTransform = colorInfo;
```

使用代码设置颜色值

`ColorTransform` 类的 `color` 属性可用于为显示对象分配具体的红、绿、蓝 (RGB) 颜色值。在下面的示例中, 当用户单击名为 `blueBtn` 的按钮时, 将使用 `color` 属性将名为 `square` 的显示对象的颜色更改为蓝色:

```
// square is a display object on the Stage.  
// blueBtn, redBtn, greenBtn, and blackBtn are buttons on the Stage.  
  
import flash.events.MouseEvent;  
import flash.geom.ColorTransform;  
  
// Get access to the ColorTransform instance associated with square.  
var colorInfo:ColorTransform = square.transform.colorTransform;  
  
// This function is called when blueBtn is clicked.  
function makeBlue(event:MouseEvent):void  
{  
    // Set the color of the ColorTransform object.  
    colorInfo.color = 0x003399;  
    // apply the change to the display object  
    square.transform.colorTransform = colorInfo;  
}  
  
blueBtn.addEventListener(MouseEvent.CLICK, makeBlue);
```

请注意，使用 `color` 属性更改显示对象的颜色时，将会完全更改整个对象的颜色，无论该对象以前是否有多 种颜色。例如，如果某个显示对象包含一个顶部有黑色文本的绿色圆，将该对象的关联 `ColorTransform` 实例的 `color` 属性设置为红色阴影时，会使整个对象（圆和文本）变为红色（因此无法再将文本与该对象的其余部分区分开来）。

使用代码更改颜色和亮度效果

假设显示对象有多种颜色（例如，数码照片），但是您不想完全重新调整对象的颜色，只想根据现有颜色来调整显示对象的颜色。这种情况下，`ColorTransform` 类包括一组可用于进行此类调整的乘数属性和偏移属性。乘数属性的名分别为 `redMultiplier`、`greenMultiplier`、`blueMultiplier` 和 `alphaMultiplier`，它们的作用像彩色照片滤镜（或彩色太阳镜）一样，可以增强或削弱显示对象上的某些颜色。偏移属性（`redOffset`、`greenOffset`、`blueOffset` 和 `alphaOffset`）可用于额外增加对象上某种颜色的值，或用于指定特定颜色可以具有的最小值。

在“属性”检查器上的“颜色”弹出菜单中选择“高级”时，这些乘数和偏移属性与 Flash 创作工具中影片剪辑元件可用的高级颜色设置相同。

下面的代码加载一个 JPEG 图像并为其应用颜色转换，当鼠标指针沿 x 轴和 y 轴移动时，将调整红色和绿色通道值。在本例中，因为未指定偏移值，所以屏幕上显示的每个颜色通道的颜色值将表示图像中原始颜色值的一个百分比，这意味着任何给定像素上显示的大部分红色或绿色都是该像素上红色或绿色的原始效果。

```

import flash.display.Loader;
import flash.events.MouseEvent;
import flash.geom.Transform;
import flash.geom.ColorTransform;
import flash.net.URLRequest;

// Load an image onto the Stage.
var loader:Loader = new Loader();
var url:URLRequest = new URLRequest("http://www.helpexamples.com/flash/images/image1.jpg");
loader.load(url);
this.addChild(loader);

// This function is called when the mouse moves over the loaded image.
function adjustColor(event:MouseEvent):void
{
    // Access the ColorTransform object for the Loader (containing the image)
    var colorTransformer:ColorTransform = loader.transform.colorTransform;

    // Set the red and green multipliers according to the mouse position.
    // The red value ranges from 0% (no red) when the cursor is at the left
    // to 100% red (normal image appearance) when the cursor is at the right.
    // The same applies to the green channel, except it's controlled by the
    // position of the mouse in the y axis.
    colorTransformer.redMultiplier = (loader.mouseX / loader.width) * 1;
    colorTransformer.greenMultiplier = (loader.mouseY / loader.height) * 1;

    // Apply the changes to the display object.
    loader.transform.colorTransform = colorTransformer;
}

loader.addEventListener(MouseEvent.MOUSE_MOVE, adjustColor);

```

旋转对象

使用 `rotation` 属性可以旋转显示对象。可以通过读取此值来了解是否旋转了某个对象，如果要旋转该对象，可以将此属性设置为一个数字（以度为单位），表示要应用于该对象的旋转量。例如，下面的代码行将名为 `square` 的对象旋转 45 度（一整周旋转的 1/8）：

```
square.rotation = 45;
```

或者，也可以使用转换矩阵来旋转显示对象，第 283 页的“[处理几何结构](#)”中对此进行了介绍。

淡化对象

可以通过控制显示对象的透明度来使显示对象部分透明（或完全透明），也可以通过更改透明度来使对象淡入或淡出。

`DisplayObject` 类的 `alpha` 属性用于定义显示对象的透明度（更确切地说是不透明度）。可以将 `alpha` 属性设置为介于 0 和 1 之间的任何值，其中 0 表示完全透明，1 表示完全不透明。例如，当使用鼠标单击名为 `myBall` 的对象时，下面的代码行将使该对象变得部分（50%）透明：

```

function fadeBall(event:MouseEvent):void
{
    myBall.alpha = .5;
}
myBall.addEventListener(MouseEvent.CLICK, fadeBall);

```

还可以使用通过 `ColorTransform` 类提供的颜色调整来更改显示对象的透明度。有关详细信息，请参阅第 251 页的“[调整 `DisplayObject` 颜色](#)”。

遮罩显示对象

可以通过将一个显示对象用作遮罩来创建一个孔洞，透过该孔洞使另一个显示对象的内容可见。

定义遮罩

若要指明一个显示对象将是另一个显示对象的遮罩，请将遮罩对象设置为被遮罩的显示对象的 `mask` 属性：

```
// Make the object maskSprite be a mask for the object mySprite.  
mySprite.mask = maskSprite;
```

被遮罩的显示对象显示在用作遮罩的显示对象的全部不透明区域之内。例如，下面的代码将创建一个包含 100 x 100 个像素的红色正方形的 `Shape` 实例和一个包含半径为 25 个像素的蓝色圆的 `Sprite` 实例。单击圆时，它被设置为正方形的遮罩，所以显示的正方形部分只是由圆完整部分覆盖的那一部分。换句话说，只有红色圆可见。

```
// This code assumes it's being run within a display object container  
// such as a MovieClip or Sprite instance.  
  
import flash.display.Shape;  
  
// Draw a square and add it to the display list.  
var square:Shape = new Shape();  
square.graphics.lineStyle(1, 0x000000);  
square.graphics.beginFill(0xffff00);  
square.graphics.drawRect(0, 0, 100, 100);  
square.graphics.endFill();  
this.addChild(square);  
  
// Draw a circle and add it to the display list.  
var circle:Sprite = new Sprite();  
circle.graphics.lineStyle(1, 0x000000);  
circle.graphics.beginFill(0x0000ff);  
circle.graphics.drawCircle(50, 50, 25);  
circle.graphics.endFill();  
this.addChild(circle);  
  
function maskSquare(event:MouseEvent):void  
{  
    square.mask = circle;  
    circle.removeEventListener(MouseEvent.CLICK, maskSquare);  
}  
  
circle.addEventListener(MouseEvent.CLICK, maskSquare);
```

用作遮罩的显示对象可拖动、设置动画，并可动态调整大小，可以在单个遮罩内使用单独的形状。遮罩显示对象不一定需要添加到显示列表中。但是，如果希望在缩放舞台时也缩放遮罩对象，或者如果希望支持用户与遮罩对象的交互（如用户控制的拖动和调整大小），则必须将遮罩对象添加到显示列表中。只要遮罩对象已添加到显示列表中，显示对象的实际 `z` 索引（从前到后的顺序）就无关紧要了。（除了显示为遮罩对象外，遮罩对象将不会出现在屏幕上。）如果遮罩对象是包含多个帧的一个 `MovieClip` 实例，则遮罩对象会沿其时间轴播放所有帧，如果没有用作遮罩对象，也会出现同样的情况。通过将 `mask` 属性设置为 `null` 可以删除遮罩：

```
// remove the mask from mySprite  
mySprite.mask = null;
```

不能使用一个遮罩对象来遮罩另一个遮罩对象。不能设置遮罩显示对象的 `alpha` 属性。只有填充可用于作为遮罩的显示对象中；笔触都会被忽略。

关于遮蔽设备字体

您可以使用显示对象遮罩用设备字体设置的文本。当使用显示对象遮罩用设备字体设置的文本时，遮罩的矩形边框会用作遮罩形状。也就是说，如果为设备字体文本创建了非矩形的显示对象遮罩，则 SWF 文件中显示的遮罩将是遮罩的矩形边框的形状，而不是遮罩本身形状。

Alpha 通道遮罩

如果遮罩显示对象和被遮罩的显示对象都使用位图缓存，则支持 Alpha 通道遮罩，如下所示：

```
// maskShape is a Shape instance which includes a gradient fill.  
mySprite.cacheAsBitmap = true;  
maskShape.cacheAsBitmap = true;  
mySprite.mask = maskShape;
```

例如，Alpha 通道遮罩的一个应用是对遮罩对象使用应用于被遮罩显示对象之外的滤镜。

在下面的示例中，将一个外部图像文件加载到舞台上。该图像（更确切地说，是加载图像的 Loader 实例）将是被遮罩的显示对象。渐变椭圆（中心为纯黑色，边缘淡变为透明）绘制在图像上；这就是 Alpha 遮罩。两个显示对象都打开了位图缓存。椭圆设置为图像的遮罩，然后使其可拖动。

```
// This code assumes it's being run within a display object container  
// such as a MovieClip or Sprite instance.  
  
import flash.display.GradientType;  
import flash.display.Loader;  
import flash.display.Sprite;  
import flash.geom.Matrix;  
import flash.net.URLRequest;  
  
// Load an image and add it to the display list.  
var loader:Loader = new Loader();  
var url:URLRequest = new URLRequest("http://www.helpexamples.com/flash/images/image1.jpg");  
loader.load(url);  
this.addChild(loader);  
  
// Create a Sprite.  
var oval:Sprite = new Sprite();  
// Draw a gradient oval.  
var colors:Array = [0x000000, 0x000000];  
var alphas:Array = [1, 0];  
var ratios:Array = [0, 255];  
var matrix:Matrix = new Matrix();  
matrix.createGradientBox(200, 100, 0, -100, -50);  
oval.graphics.beginGradientFill(GradientType.RADIAL,  
                                colors,  
                                alphas,  
                                ratios,  
                                matrix);  
oval.graphics.drawEllipse(-100, -50, 200, 100);  
oval.graphics.endFill();  
// add the Sprite to the display list  
this.addChild(oval);  
  
// Set cacheAsBitmap = true for both display objects.  
loader.cacheAsBitmap = true;  
oval.cacheAsBitmap = true;  
// Set the oval as the mask for the loader (and its child, the loaded image)  
loader.mask = oval;  
  
// Make the oval draggable.  
oval.startDrag(true);
```

对象动画

动画是使内容移动或者使内容随时间发生变化的过程。脚本动画是视频游戏的基础部分，通常用于将优美、有用的交互线索添加到其它应用程序中。

脚本动画的基本概念是变化一定要发生，而且变化一定要分时间逐步完成。使用常见的循环语句，可以很容易地在 ActionScript 中使内容重复。但是，在更新显示之前，循环将遍历其所有迭代。要创建脚本动画，需要编写 ActionScript，它随时间重复执行某个动作，每次运行时还更新屏幕。

例如，假设要创建一个简单的动画，如使球沿着屏幕运动。ActionScript 提供了一个用于跟踪时间和相应更新屏幕的简单机制，这意味着您可以编写代码，使球每次移动一点点，直到球到达目标为止。每次移动后，屏幕都会更新，从而使跨舞台的运动在查看器中可见。

从实用的观点来看，让脚本动画与 SWF 文件的帧速率同步（换句话说，每次显示或要显示新帧时都产生一个动画变化）才有意义，因为帧速率定义了 Flash Player 或 AIR 更新屏幕的频率。每个显示对象都有 `enterFrame` 事件，它根据 SWF 文件的帧速率来调度，即每帧一个事件。创建脚本动画的大多数开发人员都使用 `enterFrame` 事件作为一种方法来创建随时间重复的动作。可以编写代码以侦听 `enterFrame` 事件，每一帧都让动画球移动一定的量，当屏幕更新时（每一帧），将会在新位置重新绘制该球，从而产生了运动。

注：另一种随时间重复执行某个动作的方法是使用 `Timer` 类。每次过了指定的时间时，`Timer` 实例都会触发事件通知。可以编写通过处理 `Timer` 类的 `timer` 事件来执行动画的代码，将时间间隔设置为一个很小的值（几分之几秒）。有关使用 `Timer` 类的详细信息，请参阅第 110 页的“[控制时间间隔](#)”。

在下面的示例中，将在舞台上创建一个名为 `circle` 的圆 `Sprite` 实例。当用户单击圆时，脚本动画序列开始，从而使 `circle` 淡化（其 `alpha` 属性值减少），直到完全透明：

```
import flash.display.Sprite;
import flash.events.Event;
import flash.events.MouseEvent;

// draw a circle and add it to the display list
var circle:Sprite = new Sprite();
circle.graphics.beginFill(0x990000);
circle.graphics.drawCircle(50, 50, 50);
circle.graphics.endFill();
addChild(circle);

// When this animation starts, this function is called every frame.
// The change made by this function (updated to the screen every
// frame) is what causes the animation to occur.
function fadeCircle(event:Event):void
{
    circle.alpha -= .05;

    if (circle.alpha <= 0)
    {
        circle.removeEventListener(Event.ENTER_FRAME, fadeCircle);
    }
}

function startAnimation(event:MouseEvent):void
{
    circle.addEventListener(Event.ENTER_FRAME, fadeCircle);
}

circle.addEventListener(MouseEvent.CLICK, startAnimation);
```

当用户单击圆时，将函数 `fadeCircle()` 订阅为 `enterFrame` 事件的侦听器，这意味着每一帧都会开始调用一次该函数。通过更改 `circle` 的 `alpha` 属性，该函数会淡化圆，因此对于每个帧，圆的 `alpha` 都会减少 .05 (5%) 并会更新屏幕。最后，当 `alpha` 值为 0 (circle 完全透明) 时，`fadeCircle()` 函数作为事件侦听器将被删除，从而结束动画。

以上代码还可用来创建动画运动而不是淡化。通过用不同属性替换函数中表示 `enterFrame` 事件侦听器的 `alpha`，就可获得该属性的动画效果。例如，将以下行

```
circle.alpha -= .05;
```

更改为以下代码

```
circle.x += 5;
```

将获得 `x` 属性的动画效果，使圆移到舞台的右侧。当到达需要的 `x` 坐标时，通过更改结束动画的条件就可结束动画（即取消订阅 `enterFrame` 侦听器）。

动态加载显示内容

可以将下列任何外部显示资源加载到 ActionScript 3.0 应用程序中：

- 在 ActionScript 3.0 中创作的 SWF 文件 — 此文件可以是 `Sprite`、`MovieClip` 或扩展 `Sprite` 的任何类。
- 图像文件 — 包括 JPG、PNG 和 GIF 文件。
- AVM1 SWF 文件 — 用 ActionScript 1.0 或 2.0 编写的 SWF 文件。

使用 `Loader` 类可以加载这些资源。

加载显示对象

`Loader` 对象用于将 SWF 文件和图形文件加载到应用程序中。`Loader` 类是 `DisplayObjectContainer` 类的子类。`Loader` 对象在其显示列表中只能包含一个子显示对象，该显示对象表示它加载的 SWF 或图形文件。如下面的代码所示，在显示列表中添加 `Loader` 对象时，还可以在加载后将加载的子显示对象添加到显示列表中：

```
var pictLdr:Loader = new Loader();
var pictURL:String = "banana.jpg";
var pictURLReq:URLRequest = new URLRequest(pictURL);
pictLdr.load(pictURLReq);
this.addChild(pictLdr);
```


加载 SWF 文件或图像后，即可将加载的显示对象移到另一个显示对象容器中，如本示例中的 `container` `DisplayObjectContainer` 对象：

```
import flash.display.*;
import flash.net.URLRequest;
import flash.events.Event;
var container:Sprite = new Sprite();
addChild(container);
var pictLdr:Loader = new Loader();
var pictURL:String = "banana.jpg";
var pictURLReq:URLRequest = new URLRequest(pictURL);
pictLdr.load(pictURLReq);
pictLdr.contentLoaderInfo.addEventListener(Event.COMPLETE, imgLoaded);
function imgLoaded(event:Event):void
{
    container.addChild(pictLdr.content);
}
```

监视加载进度

文件开始加载后，就创建了 LoaderInfo 对象。LoaderInfo 对象用于提供加载进度、加载者和被加载者的 URL、媒体的字节总数及媒体的标称高度和宽度等信息。LoaderInfo 对象还调度用于监视加载进度的事件。

下图说明 LoaderInfo 对象的不同用途 — 用于 SWF 文件的主类的实例、用于 Loader 对象以及用于由 Loader 对象加载的对象：



可以将 LoaderInfo 对象作为 Loader 对象和加载的显示对象的属性进行访问。加载一开始，就可以通过 Loader 对象的 contentLoaderInfo 属性访问 LoaderInfo 对象。加载完显示对象后，也可以通过显示对象的 loaderInfo 属性，将 LoaderInfo 对象作为已加载显示对象的属性进行访问。已加载显示对象的 loaderInfo 属性是指与 Loader 对象的 contentLoaderInfo 属性相同的 LoaderInfo 对象。换句话说，LoaderInfo 对象是加载的对象与加载它的 Loader 对象之间（加载者和被加载者之间）的共享对象。

要访问加载的内容的属性，需要在 LoaderInfo 对象中添加事件侦听器，如下面的代码所示：

```

import flash.display.Loader;
import flash.display.Sprite;
import flash.events.Event;

var ldr:Loader = new Loader();
var urlReq:URLRequest = new URLRequest("Circle.swf");
ldr.load(urlReq);
ldr.contentLoaderInfo.addEventListener(Event.COMPLETE, loaded);
addChild(ldr);

function loaded(event:Event):void
{
    var content:Sprite = event.target.content;
    content.scaleX = 2;
}
  
```

有关详细信息，请参阅第 207 页的“[处理事件](#)”。

指定加载上下文

通过 Loader 类的 load() 或 loadBytes() 方法将外部文件加载到 Flash Player 或 AIR 中时，可以选择指定 context 参数。此参数是一个 LoaderContext 对象。

LoaderContext 类包括三个属性，用于定义如何使用加载的内容的上下文：

- **checkPolicyFile**: 仅当加载图像文件（不是 SWF 文件）时才会使用此属性。如果将此属性设置为 true，**Loader** 将检查策略文件的原始服务器（请参阅第 592 页的“[网站控制（策略文件）](#)”）。只有内容的来源域不是包含 **Loader** 对象的 SWF 文件所在的域时才需要此属性。如果服务器授予 **Loader** 域权限，**Loader** 域中 SWF 文件的 ActionScript 就可以访问加载图像中的数据；换句话说，可以使用 **BitmapData.draw()** 命令访问加载的图像中的数据。

请注意，来自 **Loader** 对象所在域以外的其它域的 SWF 文件可以通过调用 **Security.allowDomain()** 来允许特定的域。

- **securityDomain**: 仅当加载 SWF 文件（不是图像）时才会使用此属性。如果 SWF 文件所在的域与包含 **Loader** 对象的文件所在的域不同，则指定此属性。指定此选项时，**Flash Player** 将检查策略文件是否存在，如果存在，来自跨策略文件中允许的域的 SWF 文件可以对加载的 SWF 内容执行跨脚本操作。可以将 **flash.system.SecurityDomain.currentDomain** 指定为此参数。
- **applicationDomain**: 仅当加载使用 ActionScript 3.0 编写的 SWF 文件（不是图像或使用 ActionScript 1.0 或 2.0 编写的 SWF 文件）时才会使用此属性。加载文件时，通过将 **applicationDomain** 参数设置为 **flash.system.ApplicationDomain.currentDomain**，可以指定将该文件包括在与 **Loader** 对象相同的应用程序域中。通过将加载的 SWF 文件放在同一个应用程序域中，可以直接访问它的类。如果要加载的 SWF 文件中包含嵌入的媒体，这会很有帮助，您可以通过其关联的类名访问嵌入的媒体。有关详细信息，请参阅第 547 页的“[使用 ApplicationDomain 类](#)”。

下面的示例在从另一个域加载位图时检查策略文件：

```
var context:LoaderContext = new LoaderContext();
context.checkPolicyFile = true;
var urlReq:URLRequest = new URLRequest("http://www.[your_domain_here].com/photo11.jpg");
var ldr:Loader = new Loader();
ldr.load(urlReq, context);
```

下面的示例在从另一个域加载 SWF 时检查策略文件，以便将该文件与 **Loader** 对象放在同一个安全沙箱中。此外，该代码还将加载的 SWF 文件中的类添加到与 **Loader** 对象的类相同的应用程序域中：

```
var context:LoaderContext = new LoaderContext();
context.securityDomain = SecurityDomain.currentDomain;
context.applicationDomain = ApplicationDomain.currentDomain;
var urlReq:URLRequest = new URLRequest("http://www.[your_domain_here].com/library.swf");
var ldr:Loader = new Loader();
ldr.load(urlReq, context);
```

有关详细信息，请参阅《ActionScript 3.0 语言和组件参考》中的 [LoaderContext](#) 类。

示例：SpriteArranger

SpriteArranger 范例应用程序是在 **GeometricShapes** 范例应用程序（将单独进行介绍）的基础上构建的（请参阅第 101 页的“[示例：GeometricShapes](#)”）。

SpriteArranger 范例应用程序演示说明了处理显示对象的许多概念：

- 扩展显示对象类
- 在显示列表中添加对象
- 分层显示对象和处理显示对象容器
- 响应显示对象事件
- 使用显示对象的属性和方法

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可在文件夹 Examples/SpriteArranger 中找到 **SpriteArranger** 应用程序文件。该应用程序包含以下文件：

文件	描述
SpriteArranger.mxml 或 SpriteArranger.fla	Flash 或 Flex 中的主应用程序文件（分别为 FLA 和 MXML）。
com/example/programmingas3/SpriteArranger/CircleSprite.as	定义一种 Sprite 对象的类，该对象在屏幕上呈现为圆。
com/example/programmingas3/SpriteArranger/DrawingCanvas.as	定义画布的类，画布是包含 GeometricSprite 对象的显示对象容器。
com/example/programmingas3/SpriteArranger/SquareSprite.as	定义一种 Sprite 对象的类，该对象在屏幕上呈现为正方形。
com/example/programmingas3/SpriteArranger/TriangleSprite.as	定义一种 Sprite 对象的类，该对象在屏幕上呈现为三角形。
com/example/programmingas3/SpriteArranger/GeometricSprite.as	扩展 Sprite 对象的类，用于定义屏幕形状。 CircleSprite 、 SquareSprite 和 TriangleSprite 都扩展此类。
com/example/programmingas3/geometricshapes/IGeometricShape.as	一个基接口，用于定义要由所有几何形状类实现的方法。
com/example/programmingas3/geometricshapes/IPolygon.as	一个接口，用于定义要由具有多条边的几何形状类实现的方法。
com/example/programmingas3/geometricshapes/RegularPolygon.as	一种几何形状，这种几何形状的边长相等，并且这些边围绕形状中心对称分布。
com/example/programmingas3/geometricshapes/Circle.as	一种用于定义圆的几何形状。
com/example/programmingas3/geometricshapes/EquilateralTriangle.as	RegularPolygon 的子类，用于定义所有边长相等的三角形。
com/example/programmingas3/geometricshapes/Square.as	RegularPolygon 的子类，用于定义所有四条边相等的矩形。
com/example/programmingas3/geometricshapes/GeometricShapeFactory.as	包含“工厂方法”的一个类，用于创建给定了形状类型和尺寸的形状。

定义 **SpriteArranger** 类

用户可以使用 **SpriteArranger** 应用程序在屏幕“画布”上添加各种显示对象。

DrawingCanvas 类用于定义绘制区（一种显示对象容器），用户可以在其中添加屏幕形状。这些屏幕形状是 **GeometricSprite** 类的其中一个子类的实例。

DrawingCanvas 类

DrawingCanvas 类扩展了 **Sprite** 类，此继承是在 **DrawingCanvas** 类声明中定义的，如下所示：

```
public class DrawingCanvas extends Sprite
```

Sprite 类是 **DisplayObjectContainer** 和 **DisplayObject** 类的子类，**DrawingCanvas** 类使用这些类的方法和属性。

DrawingCanvas() 构造函数方法设置 **Rectangle** 对象 **bounds**，它是以后在绘制画布轮廓时使用的属性。然后调用 **initCanvas()** 方法，如下所示：

```
this.bounds = new Rectangle(0, 0, w, h);
initCanvas(fillColor, lineColor);
```

如下面的示例所示，**initCanvas()** 方法用于定义 **DrawingCanvas** 对象的各种属性，这些属性作为参数传递给构造函数：

```
this.lineColor = lineColor;
this.fillColor = fillColor;
this.width = 500;
this.height = 200;
```

`initCanvas()` 方法随后调用 `drawBounds()` 方法，后者使用 `DrawingCanvas` 类的 `graphics` 属性绘制画布。`graphics` 属性继承自 `Shape` 类

```
this.graphics.clear();
this.graphics.lineStyle(1.0, this.lineColor, 1.0);
this.graphics.beginFill(this.fillColor, 1.0);
this.graphics.drawRect(bounds.left - 1,
                      bounds.top - 1,
                      bounds.width + 2,
                      bounds.height + 2);

this.graphics.endFill();
```

`DrawingCanvas` 类的下列附加方法是根据用户与应用程序的交互进行调用的：

- `addShape()` 和 `describeChildren()` 方法，在第 261 页的“[在画布上添加显示对象](#)”中介绍
- `moveToBack()`、`moveDown()`、`moveToFront()` 和 `moveUp()` 方法，在第 263 页的“[重新排列显示对象层](#)”中介绍
- `onMouseUp()` 方法，在第 263 页的“[单击并拖动显示对象](#)”中介绍

GeometricSprite 类及其子类

用户在画布上可以添加的每个显示对象都是 `GeometricSprite` 类以下某个子类的实例：

- `CircleSprite`
- `SquareSprite`
- `TriangleSprite`

`GeometricSprite` 类扩展了 `flash.display.Sprite` 类：

```
public class GeometricSprite extends Sprite
```

`GeometricSprite` 类包括所有 `GeometricSprite` 对象所共有的一些属性。这些属性是根据传递到构造函数的参数，在该函数中设置的。例如：

```
this.size = size;
this.lineColor = lColor;
this.fillColor = fColor;
```

`GeometricSprite` 类的 `geometricShape` 属性用于定义 `IGeometricShape` 接口，该接口定义形状的数学属性，但不定义其可视属性。实现 `IGeometricShape` 接口的类是在 `GeometricShapes` 范例应用程序中定义的（请参阅第 101 页的“[示例：GeometricShapes](#)”）。

`GeometricSprite` 类用于定义 `drawShape()` 方法，该方法在 `GeometricSprite` 的各子类的覆盖定义中已进一步精确定义。有关详细信息，请参阅后面的“[在画布上添加显示对象](#)”一节。

`GeometricSprite` 类还提供下列方法：

- `onMouseDown()` 和 `onMouseUp()` 方法，在第 263 页的“[单击并拖动显示对象](#)”中介绍
- `showSelected()` 和 `hideSelected()` 方法，在第 263 页的“[单击并拖动显示对象](#)”中介绍

在画布上添加显示对象

当用户单击“添加形状”按钮时，应用程序将调用 `DrawingCanvas` 类的 `addShape()` 方法。它通过调用其中一个 `GeometricSprite` 子类的相应构造函数来实例化新的 `GeometricSprite`，如下面的示例所示：

```

public function addShape(shapeName:String, len:Number):void
{
    var newShape:GeometricSprite;
    switch (shapeName)
    {
        case "Triangle":
            newShape = new TriangleSprite(len);
            break;

        case "Square":
            newShape = new SquareSprite(len);
            break;

        case "Circle":
            newShape = new CircleSprite(len);
            break;
    }
    newShape.alpha = 0.8;
    this.addChild(newShape);
}

```

每个构造函数方法都调用 `drawShape()` 方法，该方法使用类（继承自 `Sprite` 类）的 `graphics` 属性来绘制相应的矢量图形。例如，`CircleSprite` 类的 `drawShape()` 方法包括下列代码：

```

this.graphics.clear();
this.graphics.lineStyle(1.0, this.lineColor, 1.0);
this.graphics.beginFill(this.fillColor, 1.0);
var radius:Number = this.size / 2;
this.graphics.drawCircle(radius, radius, radius);

```

`addShape()` 函数的倒数第二行用于设置显示对象（继承自 `DisplayObject` 类）的 `alpha` 属性，所以画布上添加的每个显示对象都有一点儿透明，这样用户就可看见它后面的对象。

`addChild()` 方法的最后一行用于在 `DrawingCanvas` 类实例的子级列表中添加新的显示对象，该实例已经在显示列表中。这样会使新的显示对象出现在舞台上。

应用程序界面上包括两个文本字段 `selectedSpriteTxt` 和 `outputTxt`。这些文本字段的文本属性由已添加到画布中或由用户选择的 `GeometricSprite` 对象的信息更新。`GeometricSprite` 类通过覆盖 `toString()` 方法来处理这种信息报告任务，如下所示：

```

public override function toString():String
{
    return this.shapeType + " of size " + this.size + " at " + this.x + ", " + this.y;
}

```

`shapeType` 属性设置为每个 `GeometricSprite` 子类的构造函数方法中的相应值。例如，`toString()` 方法可能返回最近添加到 `DrawingCanvas` 实例中的 `CircleSprite` 实例的下列值：

```
Circle of size 50 at 0, 0
```

`DrawingCanvas` 类的 `describeChildren()` 方法通过使用 `numChildren` 属性（继承自 `DisplayObjectContainer` 类）设置 `for` 循环的限制，来循环访问画布的子级列表。它会生成一个列出了每一子级的字符串，如下所示：

```

var desc:String = "";
var child:DisplayObject;
for (var i:int=0; i < this.numChildren; i++)
{
    child = this.getChildAt(i);
    desc += i + ": " + child + '\n';
}

```

所得的字符串用于设置 `outputTxt` 文本字段的 `text` 属性。

单击并拖动显示对象

当用户单击 `GeometricSprite` 实例时，应用程序将调用 `onMouseDown()` 事件处理函数。如下所示，此事件处理函数设置为侦听 `GeometricSprite` 类的构造函数中的鼠标按下事件：

```
this.addEventListener(MouseEvent.MOUSE_DOWN, onMouseDown);
```

`onMouseDown()` 方法随后调用 `GeometricSprite` 对象的 `showSelected()` 方法。如果是首次调用该对象的此方法，该方法将创建名为 `selectionIndicator` 的新 `Shape` 对象，并且使用 `Shape` 对象的 `graphics` 属性来绘制红色加亮矩形，如下所示：

```
this.selectionIndicator = new Shape();
this.selectionIndicator.graphics.lineStyle(1.0, 0xFF0000, 1.0);
this.selectionIndicator.graphics.drawRect(-1, -1, this.size + 1, this.size + 1);
this.addChild(this.selectionIndicator);
```

如果不是首次调用 `onMouseDown()` 方法，该方法仅设置 `selectionIndicator` 形状的 `visible` 属性（继承自 `DisplayObject` 类），如下所示：

```
this.selectionIndicator.visible = true;
```

`hideSelected()` 方法通过将其 `visible` 属性设置为 `false` 来隐藏以前所选对象的 `selectionIndicator` 形状。

`onMouseDown()` 事件处理函数方法还会调用 `startDrag()` 方法（继承自 `Sprite` 类），该方法包括下列代码：

```
var boundsRect:Rectangle = this.parent.getRect(this.parent);
boundsRect.width -= this.size;
boundsRect.height -= this.size;
this.startDrag(false, boundsRect);
```

这样，用户就可以在由 `boundsRect` 矩形设置的边界内在画布各处拖动选择的对象。

当用户松开鼠标按键时，将调度 `mouseUp` 事件。`DrawingCanvas` 的构造函数方法设置了下列事件侦听器：

```
this.addEventListener(MouseEvent.MOUSE_UP, onMouseUp);
```

此事件侦听器是针对 `DrawingCanvas` 对象设置的，而不是针对单个 `GeometricSprite` 对象设置的。这是因为当拖动 `GeometricSprite` 对象时，松开鼠标后它可能会在另一个显示对象（另一个 `GeometricSprite` 对象）之后结束前景中的显示对象会收到鼠标弹起事件，但用户正在拖动的显示对象则不会收到。在 `DrawingCanvas` 对象中添加侦听器可以确保始终处理该事件。

`onMouseUp()` 方法调用 `GeometricSprite` 对象的 `onMouseUp()` 方法，后者又调用 `GeometricSprite` 对象的 `stopDrag()` 方法。

重新排列显示对象层

应用程序用户界面上包括标有“后移”、“下移”、“上移”和“移到最前”的按钮。当用户单击其中一个按钮时，应用程序将调用 `DrawingCanvas` 类的相应方法：`moveToBack()`、`moveDown()`、`moveUp()` 或 `moveToFront()`。例如，`moveToBack()` 方法包括下面的代码：

```
public function moveToBack(shape:GeometricSprite):void
{
    var index:int = this.getChildIndex(shape);
    if (index > 0)
    {
        this.setChildIndex(shape, 0);
    }
}
```

该方法使用 `setChildIndex()` 方法（继承自 `DisplayObjectContainer` 类）确定显示对象位置在 `DrawingCanvas` 实例 (`this`) 的子级列表中的索引位置 0 处。

`moveDown()` 方法的作用类似，不同的是它将显示对象在 `DrawingCanvas` 实例的子级列表中的索引位置减少 1：

```
public function moveDown(shape:GeometricSprite):void
{
    var index:int = this.getChildIndex(shape);
    if (index > 0)
    {
        this.setChildIndex(shape, index - 1);
    }
}
```

moveUp() 和 moveToFront() 方法的作用与 moveToBack() 和 moveDown() 方法类似。

第 14 章：使用绘图 API

虽然导入的图像和插图非常重要，但您可以使用一项称为绘图 API 的功能（用于在 ActionScript 中绘制线条和形状）随时启动计算机中的应用程序，这就相当于一个空白画布，您可以在上面创建所需的任何图像。能够创建自己的图形可为您的应用程序提供广阔的前景。通过使用本章中介绍的方法，您可以创建绘图程序、制作交互的动画效果，或以编程方式创建您自己的用户界面元素，等等。

绘图 API 使用基础知识

使用绘图 API 简介

绘图 API 是 ActionScript 中一项内置功能的名称，您可以使用该功能来创建矢量图形（直线、曲线、形状、填充和渐变），并使用 ActionScript 在屏幕上显示它们。flash.display.Graphics 类提供了这一功能。您可以在任何 Shape、Sprite 或 MovieClip 实例中使用 ActionScript 进行绘制（使用在这每个类中定义的 graphics 属性）。（实际上，这每个类的 graphics 属性都是 Graphics 类的实例。）

如果刚刚开始学习使用代码进行绘制，可以使用 Graphics 类中包含的几种方法来简化常见形状（如圆、椭圆、矩形以及带圆角的矩形）的绘制过程。您可以将它们作为空线条或填充形状进行绘制。当您需要更高级的功能时，Graphics 类还提供了用于绘制直线和二次贝塞尔曲线的方法，您可以将这些方法与 Math 类中的三角函数配合使用，创建所需的所有形状。

Flash Player 10 和 Adobe AIR 1.5 新增了一个绘图 API，通过该 API，只需一个命令即可用编程方式绘制完整的形状。熟悉 Graphics 类以及“使用绘图 API 的基础知识”中介绍的任务后，请继续学习第 276 页的“[绘图 API 高级用法](#)”，了解有关这些绘图 API 功能的详细信息。

常见绘图 API 任务

以下是您可能需要在 ActionScript 中使用绘图 API 完成的任务，本章将对这些任务进行介绍：

- 定义线条样式和填充样式以绘制形状
- 绘制直线和曲线
- 使用方法来绘制形状（如圆、椭圆和矩形）
- 使用渐变线条和填充进行绘制
- 定义矩阵以创建渐变
- 将三角函数与绘图 API 配合使用
- 将绘图 API 与动画相结合

重要概念和术语

以下参考列表包含将会在本章中遇到的重要术语：

- 锚点：二次贝塞尔曲线的两个端点之一。
- 控制点：该点定义二次贝塞尔曲线的弯曲方向和弯曲量。弯曲的线绝不会到达控制点；但曲线就好像朝着控制点方向进行绘制的。

- 坐标空间：显示对象中所包含坐标（其子元素所在位置）的图形。
- 填充：对于带有以颜色填充的线条的形状，填充指形状内的实心部分；对于没有轮廓的形状，填充指整个形状。
- 渐变：此颜色是指从一种颜色逐渐过渡到一种或多种其它颜色，与“纯色”相对。
- 点：坐标空间中的一个位置。在 ActionScript 使用的二维坐标系中，点是按其 x 轴和 y 轴位置（点坐标）来定义的。
- 二次贝塞尔曲线：一种由特定数学公式定义的曲线类型。在这种类型的曲线中，曲线形状根据锚点（曲线端点）和控制点（定义曲线的弯曲方向和弯曲量）的位置来计算。
- 缩放：相对于原始大小的对象大小。用作动词时，对象缩放是指伸展或缩小对象以更改其大小。
- 笔触：对于带有以颜色填充的线条的形状，笔触指形状的轮廓部分；对于未填充的形状，笔触指其多个线条。
- 平移：将点坐标从一个坐标空间更改为另一个坐标空间。
- X 轴：ActionScript 使用的二维坐标系中的水平轴。
- Y 轴：ActionScript 使用的二维坐标系中的垂直轴。

完成本章中的示例

在阅读本章的过程中，您可能希望测试一些示例代码清单。因为本章介绍的是绘制可视内容，所以测试代码清单涉及运行代码以及在创建的 SWF 中查看结果。测试代码清单：

- 1 创建一个空的 Flash 文档。
- 2 在时间轴上选择一关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 使用“控制”>“测试影片”运行程序。

您将在所创建的 SWF 文件中看到代码清单的结果。

了解 Graphics 类

每个 Shape、Sprite 和 MovieClip 对象都具有一个 graphics 属性，它是 Graphics 类的一个实例。Graphics 类包含用于绘制线条、填充和形状的属性和方法。如果要将显示对象仅用作内容绘制画布，则可以使用 Shape 实例。Shape 实例的性能优于其它用于绘制的显示对象，因为它不会产生 Sprite 和 MovieClip 类中的附加功能的开销。如果希望能够在显示对象上绘制图形内容，并且还希望该对象包含其它显示对象，则可以使用 Sprite 实例。有关确定用于各种任务的显示对象的详细信息，请参阅第 240 页的“[选择 DisplayObject 子类](#)”。

绘制直线和曲线

使用 Graphics 实例进行的所有绘制均基于包含直线和曲线的基本绘制。因此，必须使用一系列相同的步骤来执行所有 ActionScript 绘制：

- 定义线条和填充样式
- 设置初始绘制位置
- 绘制直线、曲线和形状（可选择移动绘制点）
- 如有必要，完成创建填充

定义线条和填充样式

要使用 Shape、Sprite 或 MovieClip 实例的 graphics 属性进行绘制，您必须先定义在绘制时使用的样式（线条大小和颜色、填充颜色）。就像使用 Adobe® Flash® CS4 Professional 或其它绘图应用程序中的绘制工具一样，使用 ActionScript 进行绘制时，可以使用笔触进行绘制，也可以不使用笔触；可以使用填充颜色进行绘制，也可以不使用填充颜色。您可以使用 `lineStyle()` 或 `lineGradientStyle()` 方法来指定笔触的外观。要创建纯色线条，请使用 `lineStyle()` 方法。调用此方法时，您指定的最常用的值是前三个参数：线条粗细、颜色以及 Alpha。例如，该行代码指示名为 myShape 的 Shape 对象绘制 2 个像素粗、红色 (0x990000) 以及 75% 不透明的线条：

```
myShape.graphics.lineStyle(2, 0x990000, .75);
```

Alpha 参数的默认值为 1.0 (100%)，因此，如果需要完全不透明的线条，可以将该参数的值保持不变。`lineStyle()` 方法还接受两个用于像素提示和缩放模式的额外参数；有关使用这些参数的详细信息，请参阅《ActionScript 3.0 语言和组件参考》中关于 `Graphics.lineStyle()` 方法的介绍。

要创建渐变线条，请使用 `lineGradientStyle()` 方法。关于此方法的介绍请参阅第 269 页的“[创建渐变线条和填充](#)”。

如果要创建填充形状，请在开始绘制之前调用 `beginFill()`、`beginGradientFill()`、`beginBitmapFill()` 或 `beginShaderFill()` 方法。其中的最基本方法 `beginFill()` 接受以下两个参数：填充颜色以及填充颜色的 Alpha 值（可选）。例如，如果要绘制具有纯绿色填充的形状，应使用以下代码（假设在名为 myShape 的对象上进行绘制）：

```
myShape.graphics.beginFill(0x00FF00);
```

调用任何填充方法时，将隐式地结束任何以前的填充，然后再开始新的填充。调用任何指定笔触样式的方法时，将替换以前的笔触，但不会改变以前指定的填充，反之亦然。

指定了线条样式和填充属性后，下一步是指示绘制的起始点。`Graphics` 实例具有一个绘制点，就像在一张纸上的钢笔尖一样。无论绘制点位于什么位置，它都是开始执行下一个绘制动作的位置。最初，`Graphics` 对象将它绘制时所在对象的坐标空间中的点 (0, 0) 作为起始绘制点。要在其它点开始进行绘制，您可以先调用 `moveTo()` 方法，然后再调用绘制方法之一。这类似于将钢笔尖从纸上抬起，然后将其移到新位置。

确定绘制点后，可通过使用对绘制方法 `lineTo()`（用于绘制直线）和 `curveTo()`（用于绘制曲线）的一系列调用来进行绘制。

 在进行绘制时，可随时调用 `moveTo()` 方法，将绘制点移到新位置而不进行绘制。

在进行绘制时，如果已指定了填充颜色，可以指示 Adobe Flash Player 或 Adobe® AIR™ 调用 `endFill()` 方法来结束填充。如果绘制的形状没有闭合（换句话说，在调用 `endFill()` 时，绘制点不在形状的起始点），调用 `endFill()` 方法时，Flash Player 或 AIR 将自动绘制一条直线以闭合形状，该直线从当前绘制点到最近一次 `moveTo()` 调用中指定的位置。如果已开始填充并且没有调用 `endFill()`，调用 `beginFill()`（或其它填充方法之一）时，将关闭当前填充并开始新的填充。

绘制直线

调用 `lineTo()` 方法时，`Graphics` 对象将绘制一条直线，该直线从当前绘制点到指定为方法调用中的两个参数的坐标，以便使用指定的线条样式进行绘制。例如，该行代码将绘制点放在点 (100, 100) 上，然后绘制一条到点 (200, 200) 的直线：

```
myShape.graphics.moveTo(100, 100);
myShape.graphics.lineTo(200, 200);
```

以下示例绘制红色和绿色三角形，其高度为 100 个像素：

```

var triangleHeight:uint = 100;
var triangle:Shape = new Shape();

// red triangle, starting at point 0, 0
triangle.graphics.beginFill(0xFF0000);
triangle.graphics.moveTo(triangleHeight / 2, 0);
triangle.graphics.lineTo(triangleHeight, triangleHeight);
triangle.graphics.lineTo(0, triangleHeight);
triangle.graphics.lineTo(triangleHeight / 2, 0);

// green triangle, starting at point 200, 0
triangle.graphics.beginFill(0x00FF00);
triangle.graphics.moveTo(200 + triangleHeight / 2, 0);
triangle.graphics.lineTo(200 + triangleHeight, triangleHeight);
triangle.graphics.lineTo(200, triangleHeight);
triangle.graphics.lineTo(200 + triangleHeight / 2, 0);

this.addChild(triangle);

```

绘制曲线

`curveTo()` 方法可以绘制二次贝塞尔曲线。这将绘制一个连接两个点（称为锚点）的弧，同时向第三个点（称为控制点）弯曲。`Graphics` 对象使用当前绘制位置作为第一个锚点。调用 `curveTo()` 方法时，将传递以下四个参数：控制点的 x 和 y 坐标，后跟第二个锚点的 x 和 y 坐标。例如，以下代码绘制一条曲线，它从点 (100, 100) 开始，到点 (200, 200) 结束。由于控制点位于点 (175, 125)，因此，这会创建一条曲线，它先向右移动，然后向下移动：

```

myShape.graphics.moveTo(100, 100);
myShape.graphics.curveTo(175, 125, 200, 200);

```

以下示例绘制红色和绿色圆形对象，其宽度和高度均为 100 个像素。请注意，由于二次贝塞尔方程式所具有的特性，这些对象并不是完美的圆：

```

var size:uint = 100;
var roundObject:Shape = new Shape();

// red circular shape
roundObject.graphics.beginFill(0xFF0000);
roundObject.graphics.moveTo(size / 2, 0);
roundObject.graphics.curveTo(size, 0, size, size / 2);
roundObject.graphics.curveTo(size, size, size / 2, size);
roundObject.graphics.curveTo(0, size, 0, size / 2);
roundObject.graphics.curveTo(0, 0, size / 2, 0);

// green circular shape
roundObject.graphics.beginFill(0x00FF00);
roundObject.graphics.moveTo(200 + size / 2, 0);
roundObject.graphics.curveTo(200 + size, 0, 200 + size, size / 2);
roundObject.graphics.curveTo(200 + size, size, 200 + size / 2, size);
roundObject.graphics.curveTo(200, size, 200, size / 2);
roundObject.graphics.curveTo(200, 0, 200 + size / 2, 0);

this.addChild(roundObject);

```

使用内置方法绘制形状

为了便于绘制常见形状（如圆、椭圆、矩形以及带圆角的矩形），ActionScript 3.0 中提供了用于绘制这些常见形状的方法。它们是 `Graphics` 类的 `drawCircle()`、`drawEllipse()`、`drawRect()`、`drawRoundRect()` 和 `drawRoundRectComplex()` 方法。这些方法可用于替代 `lineTo()` 和 `curveTo()` 方法。但要注意，在调用这些方法之前，您仍需指定线条和填充样式。

以下示例重新创建绘制红色、绿色以及蓝色正方形的示例，其宽度和高度均为 100 个像素。以下代码使用 `drawRect()` 方法，并且还指定了填充颜色的 Alpha 为 50% (0.5)：

```
var squareSize:uint = 100;
var square:Shape = new Shape();
square.graphics.beginFill(0xFF0000, 0.5);
square.graphics.drawRect(0, 0, squareSize, squareSize);
square.graphics.beginFill(0x00FF00, 0.5);
square.graphics.drawRect(200, 0, squareSize, squareSize);
square.graphics.beginFill(0x0000FF, 0.5);
square.graphics.drawRect(400, 0, squareSize, squareSize);
square.graphics.endFill();
this.addChild(square);
```

在 `Sprite` 或 `MovieClip` 对象中，使用 `graphics` 属性创建的绘制内容始终出现在该对象包含的所有子级显示对象的后面。另外，`graphics` 属性内容不是单独的显示对象，因此，它不会出现在 `Sprite` 或 `MovieClip` 对象的子级列表中。例如，以下 `Sprite` 对象使用其 `graphics` 属性来绘制圆，并且其子级显示对象列表中包含一个 `TextField` 对象：

```
var mySprite:Sprite = new Sprite();
mySprite.graphics.beginFill(0xFFCC00);
mySprite.graphics.drawCircle(30, 30, 30);
var label:TextField = new TextField();
label.width = 200;
label.text = "They call me mellow yellow...";
label.x = 20;
label.y = 20;
mySprite.addChild(label);
this.addChild(mySprite);
```

请注意，`TextField` 将出现在使用 `graphics` 对象绘制的圆的上面。

创建渐变线条和填充

`graphics` 对象也可以绘制渐变笔触和填充，而不是纯色笔触和填充。渐变笔触是使用 `lineGradientStyle()` 方法创建的；渐变填充是使用 `beginGradientFill()` 方法创建的。

这两种方法接受相同的参数。前四个参数是必需的，即类型、颜色、Alpha 以及比率。其余四个参数是可选的，但对于高级自定义非常有用。

- 第一个参数指定要创建的渐变类型。可接受的值是 `GradientFill.LINEAR` 或 `GradientFill.RADIAL`。
- 第二个参数指定要使用的颜色值的数组。在线性渐变中，将从左向右排列颜色。在放射状渐变中，将从内到外排列颜色。数组颜色的顺序表示在渐变中绘制颜色的顺序。
- 第三个参数指定前一个参数中相应颜色的 Alpha 透明度值。
- 第四个参数指定比率或每种颜色在渐变中的重要程度。可接受的值范围是 0-255。这些值并不表示任何宽度或高度，而是表示在渐变中的位置；0 表示渐变开始，255 表示渐变结束。比率数组必须按顺序增加，并且包含的条目数与第二个和第三个参数中指定的颜色和 Alpha 数组相同。

虽然第五个参数（转换矩阵）是可选的，但通常会使用该参数，因为它提供了一种简便且有效的方法来控制渐变外观。此参数接受 `Matrix` 实例。为渐变创建 `Matrix` 对象的最简单方法是使用 `Matrix` 类的 `createGradientBox()` 方法。

定义 `Matrix` 对象以用于渐变

可以使用 `flash.display.Graphics` 类的 `beginGradientFill()` 和 `lineGradientStyle()` 方法来定义在形状中使用的渐变。定义渐变时，需要提供一个矩阵作为这些方法的其中一个参数。

定义矩阵的最简单方法是使用 **Matrix** 类的 **createGradientBox()** 方法，该方法创建一个用于定义渐变的矩阵。可以使用传递给 **createGradientBox()** 方法的参数来定义渐变的缩放、旋转和位置。**createGradientBox()** 方法接受以下参数：

- 渐变框宽度：渐变扩展到的宽度（以像素为单位）
- 渐变框高度：渐变扩展到的高度（以像素为单位）
- 渐变框旋转：将应用于渐变的旋转角度（以弧度为单位）
- 水平平移：将渐变水平移动的距离（以像素为单位）
- 垂直平移：将渐变垂直移动的距离（以像素为单位）

例如，假设渐变具有以下特性：

- **GradientType.LINEAR**
- 绿色和蓝色这两种颜色（**ratios** 数组设置为 [0, 255]）
- **SpreadMethod.PAD**
- **InterpolationMethod.LINEAR_RGB**


下面的示例显示的是几种渐变，如图所示，它们的 **createGradientBox()** 方法的 **rotation** 参数不同，但所有其它设置是相同的：

width = 100; height = 100; rotation = 0; tx = 0; ty = 0;	
width = 100; height = 100; rotation = Math.PI/4; // 45° tx = 0; ty = 0;	
width = 100; height = 100; rotation = Math.PI/2; // 90° tx = 0; ty = 0;	

下面的示例显示的是绿到蓝线性渐变的效果，如图所示，它们的 **createGradientBox()** 方法的 **rotation**、**tx** 和 **ty** 参数不同，但所有其它设置是相同的：

width = 50; height = 100; rotation = 0; tx = 0; ty = 0;	
width = 50; height = 100; rotation = 0 tx = 50; ty = 0;	
width = 100; height = 50; rotation = Math.PI/2; // 90° tx = 0; ty = 0;	
width = 100; height = 50; rotation = Math.PI/2; // 90° tx = 0; ty = 50;	

createGradientBox() 方法的 width、height、tx 和 ty 参数也会影响“放射状”渐变填充的大小和位置，如下面的示例所示：

width = 50; height = 100; rotation = 0; tx = 25; ty = 0;	
--	---

下面的代码生成了所示的最后一个放射状渐变：

```

import flash.display.Shape;
import flash.display.GradientType;
import flash.geom.Matrix;

var type:String = GradientType.RADIAL;
var colors:Array = [0x00FF00, 0x000088];
var alphas:Array = [1, 1];
var ratios:Array = [0, 255];
var spreadMethod:String = SpreadMethod.PAD;
var interp:String = InterpolationMethod.LINEAR_RGB;
var focalPtRatio:Number = 0;

var matrix:Matrix = new Matrix();
var boxWidth:Number = 50;
var boxHeight:Number = 100;
var boxRotation:Number = Math.PI/2; // 90°
var tx:Number = 25;
var ty:Number = 0;
matrix.createGradientBox(boxWidth, boxHeight, boxRotation, tx, ty);

var square:Shape = new Shape();
square.graphics.beginGradientFill(type,
    colors,
    alphas,
    ratios,
    matrix,
    spreadMethod,
    interp,
    focalPtRatio);
square.graphics.drawRect(0, 0, 100, 100);
addChild(square);

```

请注意，渐变填充的宽度和高度是由渐变矩阵的宽度和高度决定的，而不是由使用 **Graphics** 对象绘制的宽度和高度决定的。使用 **Graphics** 对象进行绘制时，您绘制的内容位于渐变矩阵中的这些坐标处。即使使用 **Graphics** 对象的形状方法之一（如 **drawRect()**），渐变也不会将其自身伸展到绘制的形状的大小；必须在渐变矩阵本身中指定渐变的大小。

下面说明了渐变矩阵的尺寸和绘图本身的尺寸之间的视觉差异：

```

var myShape:Shape = new Shape();
var gradientBoxMatrix:Matrix = new Matrix();
gradientBoxMatrix.createGradientBox(100, 40, 0, 0, 0);
myShape.graphics.beginGradientFill(GradientType.LINEAR, [0xFF0000, 0x00FF00, 0x0000FF], [1, 1, 1], [0, 128, 255], gradientBoxMatrix);
myShape.graphics.drawRect(0, 0, 50, 40);
myShape.graphics.drawRect(0, 50, 100, 40);
myShape.graphics.drawRect(0, 100, 150, 40);
myShape.graphics.endFill();
this.addChild(myShape);

```

该代码绘制三个具有相同填充样式（使用平均分布的红色、绿色和蓝色指定的）的渐变。这些渐变是使用 **drawRect()** 方法绘制的，像素宽度分别为 50、100 和 150。**beginGradientFill()** 方法中指定的渐变矩阵是使用像素宽度 100 创建的。这意味着，第一个渐变仅包含渐变色谱的一半，第二个渐变包含全部色谱，而第三个渐变包含全部色谱以及向右扩展的额外 50 蓝色像素。

lineGradientStyle() 方法的工作方式与 **beginGradientFill()** 类似，所不同的是，除了定义渐变外，您还必须在绘制之前使用 **lineStyle()** 方法指定笔触粗细。以下代码绘制一个带有红色、绿色和蓝色渐变笔触的框：

```
var myShape:Shape = new Shape();
var gradientBoxMatrix:Matrix = new Matrix();
gradientBoxMatrix.createGradientBox(200, 40, 0, 0, 0);
myShape.graphics.lineStyle(5, 0);
myShape.graphics.lineGradientStyle(GradientType.LINEAR, [0xFF0000, 0x00FF00, 0x0000FF], [1, 1, 1], [0, 128, 255], gradientBoxMatrix);
myShape.graphics.drawRect(0, 0, 200, 40);
this.addChild(myShape);
```

有关 `Matrix` 类的详细信息，请参阅第 289 页的“[使用 Matrix 对象](#)”。

将 Math 类与绘制方法配合使用

`Graphics` 对象可以绘制圆和正方形，但也可以绘制更复杂的形状，尤其是在将绘制方法与 `Math` 类的属性和方法配合使用时。`Math` 类包含人们通常很感兴趣的数学常量，如 `Math.PI`（约等于 3.14159265...），此常量表示圆的周长与其直径的比率。它还包含三角函数的方法，其中包括 `Math.sin()`、`Math.cos()` 和 `Math.tan()` 等。使用这些方法和常量绘制形状可产生更动态的视觉效果，尤其是用于重复或递归时。

`Math` 类的很多方法都要求以弧度为单位来测量圆弧，而不是使用角度。`Math` 类的一个常见用途是在这两种类型的单位之间进行转换：

```
var degrees = 121;
var radians = degrees * Math.PI / 180;
trace(radians) // 2.111848394913139
```

以下示例创建一个正弦波和余弦波以重点说明给定值的 `Math.sin()` 和 `Math.cos()` 方法之间的差异。

```
var sinWavePosition = 100;
var cosWavePosition = 200;
var sinWaveColor:uint = 0xFF0000;
var cosWaveColor:uint = 0x00FF00;
var waveMultiplier:Number = 10;
var waveStretcher:Number = 5;

var i:uint;
for(i = 1; i < stage.stageWidth; i++)
{
    var sinPosY:Number = Math.sin(i / waveStretcher) * waveMultiplier;
    var cosPosY:Number = Math.cos(i / waveStretcher) * waveMultiplier;

    graphics.beginFill(sinWaveColor);
    graphics.drawRect(i, sinWavePosition + sinPosY, 2, 2);
    graphics.beginFill(cosWaveColor);
    graphics.drawRect(i, cosWavePosition + cosPosY, 2, 2);
}
```

使用绘图 API 进行动画处理

使用绘图 API 创建内容的一个优点是，您并不限于将内容放置一次。可通过保留和修改用于绘制的变量来修改所绘制的内容。您可以通过更改变量和重绘（在一段帧上或使用计时器）来利用原有的动画。

例如，以下代码更改每个经过的帧（通过侦听 `Event.ENTER_FRAME` 事件）的显示内容以增加当前度数，指示 `graphics` 对象清除内容并在更新位置进行重绘。

```
stage.frameRate = 31;

var currentDegrees:Number = 0;
var radius:Number = 40;
var satelliteRadius:Number = 6;

var container:Sprite = new Sprite();
container.x = stage.stageWidth / 2;
container.y = stage.stageHeight / 2;
addChild(container);
var satellite:Shape = new Shape();
container.addChild(satellite);

addEventListener(Event.ENTER_FRAME, doEveryFrame);

function doEveryFrame(event:Event):void
{
    currentDegrees += 4;
    var radians:Number = getRadians(currentDegrees);
    var posX:Number = Math.sin(radians) * radius;
    var posY:Number = Math.cos(radians) * radius;
    satellite.graphics.clear();
    satellite.graphics.beginFill(0);
    satellite.graphics.drawCircle(posX, posY, satelliteRadius);
}
function getRadians(degrees:Number):Number
{
    return degrees * Math.PI / 180;
}
```

要产生明显不同的效果，您可以尝试修改代码开头的初始种子变量（`currentDegrees`、`radius` 和 `satelliteRadius`）。例如，尝试缩小 `radius` 变量和 / 或增大 `totalSatellites` 变量。这只是一个说明绘图 API 如何创建可视显示内容（其复杂性掩盖了创建简便性）的示例。

示例：Algorithmic Visual Generator

Algorithmic Visual Generator 示例在舞台上动态绘制几个“卫星”，即在圆形轨道中运动的圆形物。要阐述的功能包括：

- 使用绘图 API 绘制具有动态外观的基本形状
- 将用户交互与绘图中使用的属性相关联
- 清除每个帧中的舞台内容并重绘以利用原有的动画

上一小节中的示例使用 `Event.ENTER_FRAME` 事件对唯一的“卫星”进行动画处理。该示例在此基础之上进一步扩展，生成一个包含一系列滑块的控制面板，这些滑块会立即更新若干卫星的可视显示内容。此示例将代码格式设置为外部类，并将卫星创建代码包装在循环中，以将对每个卫星的引用存储在 `satellites` 数组中。

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可以在 `Samples/AlgorithmicVisualGenerator` 文件夹中找到应用程序文件。此文件夹包含以下文件：

文件	描述
AlgorithmicVisualGenerator.fla	Flash 中的主应用程序文件 (FLA)。
com/example/programmingas3/algorithmic/AlgorithmicVisualGenerator.as	此类提供应用程序的主要功能，其中包括在舞台上绘制卫星，以及从控制面板中响应事件以更新影响卫星绘制的变量。
com/example/programmingas3/algorithmic/ControlPanel.as	此类管理用户与几个滑块之间的交互并在发生此类交互时调度事件。
com/example/programmingas3/algorithmic/Satellite.as	此类表示在轨道中围绕中心点旋转的显示对象，并包含与其当前绘制状态有关的属性。

设置侦听器

应用程序先创建三个侦听器。第一个侦听器侦听从控制面板中调度的事件：必须对卫星进行重新构建。第二个侦听器侦听对 SWF 文件的舞台大小的更改。第三个侦听器侦听 SWF 文件中每个经过的帧，并使用 `doEveryFrame()` 函数进行重绘。

创建卫星

在设置这些侦听器后，将调用 `build()` 函数。此函数先调用 `clear()` 函数，后者清空 `satellites` 数组，并清除以前在舞台上绘制的任何内容。这是必要的，因为每当控制面板发送事件来执行此操作时，都可能会重新调用 `build()` 函数，例如在颜色设置已发生变化时。在这种情况下，必须删除并重新创建卫星。

该函数随后创建一些卫星，并设置创建所需的初始属性，如 `position` 变量（从轨道中的随机位置开始）和 `color` 变量（在本示例中，该变量在创建卫星后不会发生改变）。

创建每个卫星时，会将对它的引用添加到 `satellites` 数组中。调用 `doEveryFrame()` 函数时，它将更新此数组中的所有卫星。

更新卫星位置

`doEveryFrame()` 函数是应用程序动画过程的核心。将为每个帧调用该函数，频率等于 SWF 文件的帧频。由于绘制变量略微发生了变化，因此，这会利用原有的动画外观。

该函数先清除以前绘制的所有内容，然后再重绘背景。接下来，它循环访问每个卫星容器，增加每个卫星的 `position` 属性以及更新 `radius` 和 `orbitRadius` 属性（这些属性可能由于用户与控制面板的交互而发生了变化）。最后，通过调用 `Satellite` 类的 `draw()` 方法，在新位置对卫星进行更新。

请注意，计数器 `i` 最多只增加到 `visibleSatellites` 变量。这是因为，如果用户通过控制面板限制了显示的卫星数，则不会重绘循环中的其余卫星，而应将其隐藏起来。这种情况发生在紧靠负责绘制的循环后面的循环中。

当 `doEveryFrame()` 函数完成时，将在屏幕上的位置中更新 `visibleSatellites` 数量。

响应用户交互

用户交互是通过控制面板发生的，它是由 `ControlPanel` 类管理的。此类设置一个侦听器，并为每个滑块设置单独的最小、最大和默认值。当用户移动这些滑块时，将调用 `changeSetting()` 函数。此函数更新控制面板的属性。如果更改需要重新构建显示内容，则会调度一个事件，随后将在主应用程序文件中对该事件进行处理。当控制面板设置发生变化时，`doEveryFrame()` 函数将使用更新的变量绘制每个卫星。

进一步自定义

本示例只是概要介绍了使用绘图 API 生成可视内容方面的基础知识。它使用相对较少的几行代码来创建似乎非常复杂的交互式体验。尽管如此，您还是可以对本示例进行较小的改动以进行扩展。下面列出了一些很好的方法：

- `doEveryFrame()` 函数可以增加卫星的颜色值。
- `doEveryFrame()` 函数可能会随着时间的推移缩小或增大卫星半径。
- 卫星并不一定是圆形的，例如，可以使用 `Math` 类根据正弦波来移动其半径。
- 卫星可以使用碰撞检测来检查是否与其它卫星重叠。

可以将绘图 API 作为在 Flash 创作环境中创建视觉效果的替代方法，以便在运行时绘制基本形状。但是，它也可以用来创建涵盖范围很广且无法手动创建的各种视觉效果。通过使用绘图 API 和一些数学函数，ActionScript 作者可能实现很多意想不到的创作效果。

绘图 API 高级用法

使用高级绘图 API 简介

Flash Player 10 和 Adobe AIR 1.5 引入了对一组高级绘图功能的支持。绘图 API 增强功能对早期版本中的绘图方法进行了扩展，因此通过建立数据集，即可生成形状、在运行时更改形状及创建三维效果。绘图 API 增强功能将现有方法合并为其它命令。这些命令利用矢量数组和枚举类为绘图方法提供数据集。使用矢量数组，可以快速呈现更复杂的形状，开发人员可以用编程方式为运行时呈现的动态形状更改数组值。

下面几节介绍 Flash Player 10 中引入的绘制功能：第 277 页的“[绘制路径](#)”、第 278 页的“[定义缠绕规则](#)”、第 280 页的“[使用图形数据类](#)”和第 282 页的“[关于使用 drawTriangles\(\)](#)”。

常见高级绘图 API 任务

您可能需要使用 ActionScript 中的高级绘图 API 来完成以下任务：

- 使用 `Vector` 对象存储绘制方法的数据
- 以编程方式定义用于绘制形状的路径
- 定义缠绕规则，以确定如何填充重叠形状
- 使用图形数据类
- 使用三角形和绘制方法实现三维效果

重要概念和术语

以下参考列表包含本节中会遇到的重要术语：

- 矢量：数据类型完全相同的值组成的数组。`Vector` 对象可存储绘制方法使用单个命令构建线条和形状时所用值的数组。有关 `Vector` 对象的详细信息，请参阅第 129 页的“[索引数组](#)”。
- 路径：路径由一条或多条直线段或曲线段组成。每个线段的起点和终点都由坐标标记，就像用于固定线的针。路径可以是闭合的，例如圆；也可以是开放的，且具有不同的端点，例如波浪线。
- 缠绕：由渲染器解释的路径方向，包括正向（顺时针）或负向（逆时针）。
- `GraphicsStroke`：用于设置线条样式的类。虽然“笔触”并非绘图 API 增强功能中的术语，但使用类以线条样式自身的填充属性来指定该线条样式却是这些增强功能的一部分。您可以使用 `GraphicsStroke` 类动态调整线条的样式。

- Fill 对象：使用 `flash.display.GraphicsBitmapFill` 和 `flash.display.GraphicsGradientFill` 等传递给绘图命令 `Graphics.drawGraphicsData()` 的显示类所创建的对象。Fill 对象和增强的绘图命令引入了一种面向对象程度更高的编程方法，用于复现 `Graphics.beginBitmapFill()` 和 `Graphics.beginGradientFill()` 的效果。

绘制路径

有关绘制线条和曲线的节中（请参阅第 266 页的“[绘制直线和曲线](#)”）介绍了一些命令，它们可用于绘制单个线条（`Graphics.lineTo()`）或曲线（`Graphics.curveTo()`），然后将得到的线移至另一点（`Graphics.moveTo()`），从而组成形状。Flash Player 10 和 Adobe AIR 1.5 引入了对 ActionScript 绘图 API 增强功能（如 `Graphics.drawPath()` 和 `Graphics.drawTriangles()`，这些方法采用现有的绘图命令作为参数）的支持。因此，可以在单个语句中运行一系列 `Graphics.lineTo()`、`Graphics.curveTo()` 或 `Graphics.moveTo()` 命令。

其中的两项绘图 API 增强功能使 `Graphics.drawPath()` 和 `Graphics.drawTriangles()` 可以合并现有命令：

- `GraphicsPathCommand` 枚举类：`GraphicsPathCommand` 类将多个绘制命令与常数值进行关联。可以使用一系列这种值作为 `Graphics.drawPath()` 方法的参数。这样，通过单个命令可以呈现整个形状或多个形状。此外，还可以动态更改传递给这些方法的值，以更改现有形状。
- 矢量数组：矢量数组包含特定数据类型的一系列值。因此，将一系列 `GraphicsPathCommand` 常量存储在一个 `Vector` 对象中，将一系列坐标存储在另一个 `Vector` 对象中。`Graphics.drawPath()` 或 `Graphics.drawTriangles()` 共同指派这些值，以生成绘制路径或形状。

不再需要针对每个形状片段使用单独的命令。例如，`Graphics.drawPath()` 方法将 `Graphics.moveTo()`、`Graphics.lineTo()` 和 `Graphics.curveTo()` 并入一个方法中。无需单独调用每个方法，它们已抽象到 `GraphicsPathCommand` 类定义的数字标识符中。`moveTo()` 操作由 1 表示，`lineTo()` 操作由 2 表示。这些值的数组存储在 `Vector.<int>` 对象中，供 `commands` 参数中使用。然后，在 `Vector.<Number>` 对象中为 `data` 参数创建包含坐标的另一个数组。每个 `GraphicsPathCommand` 值都与 `data` 参数中存储的坐标值相对应，其中两个连续数字定义目标坐标空间中的位置。

注：矢量中的值不是 `Point` 对象；该矢量是一系列数字，其中由两个数字组成的每个组表示一个 x/y 坐标对。

`Graphics.drawPath()` 方法将每个命令与其各自的点值（由两个或四个数字组成的集合）相匹配，以在 `Graphics` 对象中生成路径：

```

package{
import flash.display.*;

public class DrawPathExample extends Sprite {
    public function DrawPathExample(){

        var square_commands:Vector.<int> = new Vector.<int>(5,true);
        square_commands[0] = 1;//moveTo
        square_commands[1] = 2;//lineTo
        square_commands[2] = 2;
        square_commands[3] = 2;
        square_commands[4] = 2;


        var square_coord:Vector.<Number> = new Vector.<Number>(10,true);
        square_coord[0] = 20; //x
        square_coord[1] = 10; //y
        square_coord[2] = 50;
        square_coord[3] = 10;
        square_coord[4] = 50;
        square_coord[5] = 40;
        square_coord[6] = 20;
        square_coord[7] = 40;
        square_coord[8] = 20;
        square_coord[9] = 10;

        graphics.beginFill(0x442266);//set the color
        graphics.drawPath(square_commands, square_coord);
    }
}
}

```

定义缠绕规则

Flash Player 10 和 Adobe AIR 1.5 还引入了路径“缠绕”（即路径的方向）这一概念。路径的缠绕可以是正向的（顺时针），也可以是负向的（逆时针）。渲染器为 **data** 参数解释矢量所提供坐标的顺序确定了缠绕的方向。



正向缠绕和负向缠绕

A. 指示绘制方向的箭头 **B.** 正向缠绕（顺时针） **C.** 负向缠绕（逆时针）


此外，请注意 `Graphics.drawPath()` 方法可选的第三个参数“winding”：

```
drawPath(commands:Vector.<int>, data:Vector.<Number>, winding:String = "evenOdd") :void
```

在这种情况下，第三个参数是一个字符串或常量，用于指定相交路径的缠绕或填充规则。（这些常数值在 `GraphicsPathWinding` 类中定义为 `GraphicsPathWinding.EVEN_ODD` 或 `GraphicsPathWinding.NON_ZERO`。）当路径相交时，缠绕规则十分重要。

奇偶规则是标准的缠绕规则，早期的绘图 API 都使用此规则。奇偶规则也是 `Graphics.drawPath()` 方法的默认规则。使用奇偶缠绕规则时，任何相交路径都交替使用开放填充与闭合填充。如果使用同一填充绘制的两个正方形相交，则不会填充相交的区域。通常，相邻区域不会都填充或都不填充。


另一方面，非零规则依靠缠绕（绘制方向）来确定是否填充相交路径定义的区域。当相对缠绕的路径相交时，不填充所定义的区域，这与奇偶规则十分类似。对于相同缠绕的路径，将填充本来不填充的区域：



用于相交区域的缠绕规则
A. 奇偶缠绕规则 B. 非零缠绕规则

缠绕规则名称

这些名称系指用于定义如何管理填充的更具体规则。正向缠绕路径将得到赋值 `+1`；负向缠绕路径将得到赋值 `-1`。以形状上闭合区域中的一点为起点，绘制一个从该点向外无限延伸的线条。使用该线条与路径相交的次数以及这些路径的组合值来确定填充。对于奇偶缠绕，使用该线条与路径相交的次数。如果计数为奇数，则填充相交区域。如果计数为偶数，则不填充相交区域。对于非零缠绕，使用赋予路径的值。如果路径的组合值不为 `0`，则填充相交区域。如果组合值为 `0`，则不填充相交区域。




缠绕规则计数和填充
A. 奇偶缠绕规则 B. 非零缠绕规则

使用缠绕规则

这些填充规则很复杂，但有些情况下必须使用它们。例如绘制星形时。如果使用标准奇偶规则，该形状需要十个不同的线条。如果使用非零缠绕规则，所需十个线条将减少为五个。下面的 ActionScript 使用五个线条和非零缠绕规则来绘制星形：

```
fill.graphics.beginFill(0x60A0FF);graphics.drawPath( Vector.<int>([1,2,2,2,2]), Vector.<Number>([66,10,23,127, 122,50, 10,49, 109,127]), GraphicsPathWinding.NON_ZERO);
```

星形如下所示：



使用不同缠绕规则的星形
A.奇偶 10 个线条 **B.**奇偶 5 个线条 **C.**非零 5 个线条

如果对图像进行了动画处理，或图像用作三维对象上的纹理且发生重叠，则缠绕规则会变得更为重要。

使用图形数据类

Flash Player 10 和 Adobe AIR 1.5 在 [IGraphicsData](#) 类型（每个类所实现的接口）的 `flash.display` 包中引入了类的集合。实现 `IGraphicsData` 接口的类充当绘图 API 的方法的数据容器。

下面的类实现 `IGraphicsData` 接口：

- `GraphicsBitmapFill`
- `GraphicsEndFill`
- `GraphicsGradientFill`
- `GraphicsPath`
- `GraphicsShaderFill`
- `GraphicsSolidFill`
- `GraphicsStroke`
- `GraphicsTrianglePath`

通过这些类，可以将全部绘制存储在 `IGraphicsData` 类型的矢量对象数组 (`Vector.<IGraphicsData>`) 中，该数组可以重复用作其它形状实例的数据源，也可以存储绘制信息供以后使用。

请注意，每个填充样式有多个填充类，但只有一个笔触类。ActionScript 只有一个笔触类 `IGraphicsData`，因为该笔触类使用填充类来定义自己的样式。因此，每个笔触实际上是笔触类和填充类。否则，这些图形数据类的 API 会镜像它们在 `flash.display.Graphics` 类中表示的方法：

<code>Graphics</code> 方法	<code>Data</code> 类
<code>beginBitmapFill()</code>	<code>GraphicsBitmapFill</code>
<code>beginFill()</code>	<code>GraphicsSolidFill</code>
<code>beginGradientFill()</code>	<code>GraphicsGradientFill</code>
<code>beginShaderFill()</code>	<code>GraphicsShaderFill</code>
<code>lineBitmapStyle()</code>	<code>GraphicsStroke + GraphicsBitmapFill</code>
<code>lineGradientStyle()</code>	<code>GraphicsStroke + GraphicsGradientFill</code>
<code>lineShaderStyle()</code>	<code>GraphicsStroke + GraphicsShaderFill</code>

Graphics 方法	Data 类
lineStyle()	GraphicsStroke + GraphicsSolidFill
moveTo()	GraphicsPath
lineTo()	
curveTo()	
drawPath()	
drawTriangles()	GraphicsTrianglePath

此外，[GraphicsPath](#) 类有自己的 `GraphicsPath.moveTo()`、`GraphicsPath.lineTo()`、`GraphicsPath.curveTo()`、`GraphicsPath.wideLineTo()` 和 `GraphicsPath.wideMoveTo()` 实用程序方法，供轻松地为 `GraphicsPath` 实例定义这些命令。这些实用程序方法简化了直接定义或更新这些命令和数据值的过程。

一旦获得 `IGraphicsData` 实例的集合，即可使用 `Graphics.drawGraphicsData()` 方法来呈现图形。`Graphics.drawGraphicsData()` 方法通过绘图 API 按顺序运行各个 `IGraphicsData` 实例的矢量：

```
// stroke object
var stroke:GraphicsStroke = new GraphicsStroke(3);
stroke.joints = JointStyle.MITER;
stroke.fill = new GraphicsSolidFill(0x102020); // solid stroke

// fill object
var fill:GraphicsGradientFill = new GraphicsGradientFill();
fill.colors = [0x0000FF, 0xEEFFEE];
fill.matrix = new Matrix();
fill.matrix.createGradientBox(70, 70, Math.PI/2);
// path object
var path:GraphicsPath = new GraphicsPath(new Vector.<int>(), new Vector.<Number>());
path.commands.push(1, 2, 2);
path.data.push(125, 0, 50, 100, 175, 0);

// combine objects for complete drawing
var drawing:Vector.<IGraphicsData> = new Vector.<IGraphicsData>();
drawing.push(stroke, fill, path);

// draw the drawing
graphics.drawGraphicsData(drawing);
```

通过修改示例中绘制所用路径的一个值，可以多次重绘形状，以生成更复杂的图像：

```
// draw the drawing multiple times
// change one value to modify each variation
graphics.drawGraphicsData(drawing);
path.data[2] += 200;
graphics.drawGraphicsData(drawing);
path.data[2] -= 150;
graphics.drawGraphicsData(drawing);
path.data[2] += 100;
graphics.drawGraphicsData(drawing);
path.data[2] -= 50; graphicsS.drawGraphicsData(drawing);
```

尽管 `IGraphicsData` 对象可以定义填充和笔触样式，但这两种样式并不是必需的。换句话说，可以在使用 `IGraphicsData` 对象绘制已保存路径集合的同时，使用 `Graphics` 类方法设置样式，反之亦然。

注：开始新的绘制之前，请使用 `Graphics.clear()` 方法清除以前的绘制；除非如上例所示，要在原始绘制的基础上继续绘制。在更改路径或 `IGraphicsData` 对象集合的某个部分时，请重绘整个绘制来查看所做的更改。

使用图形数据类时，只要绘制了三点或更多点，就会呈现填充，这是因为形状实际上在该点闭合。即使填充闭合，笔触也不会闭合；这一行为与使用多个 `Graphics.lineTo()` 或 `Graphics.moveTo()` 命令时不同。

关于使用 `drawTriangles()`

Flash Player 10 和 Adobe AIR 1.5 中引入的另一个高级方法是 `Graphics.drawTriangles()`，它与 `Graphics.drawPath()` 方法类似。`Graphics.drawTriangles()` 方法还使用 `Vector.<Number>` 对象指定用于绘制路径的点位置。`<Number>` 对象来指定用于绘制路径的点位置。

但是，`Graphics.drawTriangles()` 方法的实际作用是便于通过 ActionScript 实现三维效果。有关使用 `Graphics.drawTriangles()` 生成三维效果的信息，请参阅第 433 页的“[通过三角形获得 3D 效果](#)”。

第 15 章：处理几何结构

`flash.geom` 包中包含用于定义几何对象（如，点、矩形和转换矩阵）的类。您可使用这些类来定义在其它类中使用的对象的属性。

几何学基础知识

处理几何学简介

很多人都可能会在学校努力学习几何学这门学科，过后又几乎忘记殆尽；但就算只了解一点这方面的知识，也可能会在 ActionScript 中派上大用场。

`flash.geom` 包中包含用于定义几何对象（如，点、矩形和转换矩阵）的类。这些类本身并不一定提供功能，但它们用于定义在其它类中使用的对象的属性。

所有几何类都基于以下概念：将屏幕上的位置表示为二维平面。可以将屏幕看作是具有水平 (`x`) 轴和垂直 (`y`) 轴的平面图形。屏幕上的任何位置（或“点”）可以表示为 `x` 和 `y` 值对，即该位置的“坐标”。

每个显示对象（包括舞台）都具有自己的“坐标空间”；实质上，这是其用于标绘子显示对象、图画等位置的图形。通常，“原点”（`x` 和 `y` 轴相交的位置，其坐标为 `0, 0`）位于显示对象的左上角。尽管这始终适用于舞台，但并不一定适用于任何其它显示对象。正如在标准二维坐标系中一样，`x` 轴上的值越往右越大，越往左越小；对于原点左侧的位置，`x` 坐标为负值。但是，与传统的坐标系相反，在 ActionScript 中，屏幕 `y` 轴上的值越往下越大，越往上越小（原点上面的 `y` 坐标为负值）。由于舞台左上角是其坐标空间的原点，因此，舞台上的任何对象的 `x` 坐标大于 `0` 并小于舞台宽度，`y` 坐标大于 `0` 并小于舞台高度。

可以使用 `Point` 类实例来表示坐标空间中的各个点。您可以创建一个 `Rectangle` 实例来表示坐标空间中的矩形区域。对于高级用户，可以使用 `Matrix` 实例将多个或复杂变形应用于显示对象。通过使用显示对象的属性，可以将很多简单变形（如旋转、位置以及缩放变化）直接应用于该对象。有关使用显示对象属性应用变形的详细信息，请参阅第 241 页的“[处理显示对象](#)”。

常见几何学任务

您可能需要使用 ActionScript 中的几何类来完成以下任务：

- 计算两个点之间的距离
- 确定不同坐标空间中的点坐标
- 使用角度和距离移动显示对象
- 处理 `Rectangle` 实例：
 - 重新定位 `Rectangle` 实例
 - 调整 `Rectangle` 实例的大小
 - 确定 `Rectangle` 实例的组合大小或重叠区域
- 创建 `Matrix` 对象
- 使用 `Matrix` 对象将变形应用于显示对象

重要概念和术语

以下参考列表包含将会在本章中遇到的重要术语：

- 笛卡尔坐标：通常，坐标采用一对数字的形式（如 5, 12 或 17, -23）。两个数字分别是 x 坐标和 y 坐标。
- 坐标空间：显示对象中包含的坐标（其子元素所在的位置）的图形。
- 原点：坐标空间中的一个点，x 轴和 y 轴在此位置相交。该点的坐标为 0, 0。
- 点：坐标空间中的一个位置。在 ActionScript 使用的二维坐标系中，点是按其 x 轴和 y 轴位置（点坐标）来定义的。
- 注册点：显示对象的坐标空间的原点（0, 0 坐标）。
- 缩放：相对于原始大小的对象大小。用作动词时，对象缩放是指伸展或缩小对象以更改其大小。
- 平移：将点的坐标从一个坐标空间更改为另一个坐标空间。
- 变形：对图形的可视特性进行的调整，如旋转对象、改变其缩放比例、倾斜或扭曲其形状或者改变其颜色。
- X 轴：ActionScript 使用的二维坐标系中的水平轴。
- Y 轴：ActionScript 使用的二维坐标系中的垂直轴。

完成本章中的示例

本章中的许多示例都说明了计算或更改值；这些示例中大部分都包括相应的 trace() 函数调用以说明代码的结果。要测试这些示例，请执行以下操作：

- 1 使用 Flash 创作工具创建空白文档
- 2 在时间轴上选择一关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 通过“控制”>“测试影片”运行程序。

您将在“输出”面板中看到该代码清单的 trace() 函数的结果。

本章中某些示例说明了如何将变形应用于显示对象。对于这些示例而言，示例的结果可直观地看到而不是通过文本输出看到。要测试变形示例，请执行以下操作：

- 1 使用 Flash 创作工具创建空白文档
- 2 在时间轴上选择一关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 在舞台上创建一个影片剪辑元件实例。例如，绘制一个形状，选中它，选择“修改”>“转换为元件”，并给元件指定一个名称。
- 5 在“属性”检查器中选中舞台影片剪辑，为实例指定一个实例名称。该名称应与用于显示示例代码清单中的对象的名称相匹配，例如，如果代码清单将变形应用于名为 myDisplayObject 的对象，则应将影片剪辑实例的名称也命名为 myDisplayObject。
- 6 通过“控制”>“测试影片”运行程序。

在屏幕上，您将看到将变形应用于代码清单中指定的对象后的结果。

用于测试示例代码清单的方法在第 29 页的“[测试本章内的示例代码清单](#)”中有更详细的介绍。

使用 Point 对象

Point 对象定义一对笛卡尔坐标。它表示二维坐标系中的某个位置。其中 x 表示水平轴，y 表示垂直轴。

要定义 Point 对象, 请设置它的 x 和 y 属性, 如下所示:

```
import flash.geom.*;
var pt1:Point = new Point(10, 20); // x == 10; y == 20
var pt2:Point = new Point();
pt2.x = 10;
pt2.y = 20;
```

确定两点之间的距离

可以使用 Point 类的 distance() 方法确定坐标空间两点之间的距离。例如, 下面的代码确定同一显示对象容器中两个显示对象 (circle1 和 circle2) 的注册点之间的距离:

```
import flash.geom.*;
var pt1:Point = new Point(circle1.x, circle1.y);
var pt2:Point = new Point(circle2.x, circle2.y);
var distance:Number = Point.distance(pt1, pt2);
```

平移坐标空间

如果两个显示对象位于不同的显示对象容器中, 则它们可能位于不同的坐标空间。您可以使用 DisplayObject 类的 localToGlobal() 方法将坐标平移到舞台中相同 (全局) 坐标空间。例如, 下面的代码确定不同显示对象容器中两个显示对象 (circle1 和 circle2) 的注册点之间的距离:

```
import flash.geom.*;
var pt1:Point = new Point(circle1.x, circle1.y);
pt1 = circle1.localToGlobal(pt1);
var pt2:Point = new Point(circle2.x, circle2.y);
pt2 = circle2.localToGlobal(pt2);
var distance:Number = Point.distance(pt1, pt2);
```

同样, 要确定名为 target 的显示对象的注册点与舞台上特定点之间的距离, 您可以使用 DisplayObject 类的 localToGlobal() 方法:

```
import flash.geom.*;
var stageCenter:Point = new Point();
stageCenter.x = this.stage.stageWidth / 2;
stageCenter.y = this.stage.stageHeight / 2;
var targetCenter:Point = new Point(target.x, target.y);
targetCenter = target.localToGlobal(targetCenter);
var distance:Number = Point.distance(stageCenter, targetCenter);
```

按指定的角度和距离移动显示对象

您可以使用 Point 类的 polar() 方法将显示对象按特定角度移动特定距离。例如, 下列代码按 60 度将 myDisplayObject 对象移动 100 个像素:

```
import flash.geom.*;
var distance:Number = 100;
var angle:Number = 2 * Math.PI * (90 / 360);
var translatePoint:Point = Point.polar(distance, angle);
myDisplayObject.x += translatePoint.x;
myDisplayObject.y += translatePoint.y;
```

Point 类的其它用法

您可以将 Point 对象用于以下方法和属性:

类	方法或属性	描述
DisplayObjectContainer	areInaccessibleObjectsUnderPoint() getObjectsUnderPoint() int()	用于返回显示对象容器中某个点下的对象的列表。
BitmapData	hitTest()	用于定义 BitmapData 对象中的像素以及要检查点击的点。
BitmapData	applyFilter() copyChannel() merge() paletteMap() pixelDissolve() threshold()	用于定义那些定义操作的矩形的位置。
Matrix	deltaTransformPoint() transformPoint()	用于定义您要对其应用变形的点。
Rectangle	bottomRight 大小 topLeft	用于定义这些属性。

使用 Rectangle 对象

Rectangle 对象定义一个矩形区域。Rectangle 对象有一个位置，该位置由其左上角的 x 和 y 坐标以及 width 属性和 height 属性定义。通过调用 Rectangle() 构造函数可以定义新 Rectangle 对象的这些属性，如下所示：

```
import flash.geom.Rectangle;
var rx:Number = 0;
var ry:Number = 0;
var rwidth:Number = 100;
var rheight:Number = 50;
var rect1:Rectangle = new Rectangle(rx, ry, rwidth, rheight);
```

调整 Rectangle 对象的大小和进行重新定位

有多种方法调整 Rectangle 对象的大小和进行重新定位。

您可以通过更改 Rectangle 对象的 x 和 y 属性直接重新定位该对象。这对 Rectangle 对象的宽度或高度没有任何影响。

```
import flash.geom.Rectangle;
var x1:Number = 0;
var y1:Number = 0;
var width1:Number = 100;
var height1:Number = 50;
var rect1:Rectangle = new Rectangle(x1, y1, width1, height1);
trace(rect1) // (x=0, y=0, w=100, h=50)
rect1.x = 20;
rect1.y = 30;
trace(rect1); // (x=20, y=30, w=100, h=50)
```

如以下代码所示，如果更改 Rectangle 对象的 left 或 top 属性，也可以重新定位，并且该对象的 x 和 y 属性分别与 left 和 top 属性匹配。但是，Rectangle 对象的左下角位置不发生更改，所以调整了对象的大小。

```

import flash.geom.Rectangle;
var x1:Number = 0;
var y1:Number = 0;
var width1:Number = 100;
var height1:Number = 50;
var rect1:Rectangle = new Rectangle(x1, y1, width1, height1);
trace(rect1) // (x=0, y=0, w=100, h=50)
rect1.left = 20;
rect1.top = 30;
trace(rect1); // (x=20, y=30, w=80, h=20)

```

同样，如下面的示例所示，如果更改 `Rectangle` 对象的 `bottom` 或 `right` 属性，该对象的左上角位置不发生更改，所以相应地调整了对象的大小。

```

import flash.geom.Rectangle;
var x1:Number = 0;
var y1:Number = 0;
var width1:Number = 100;
var height1:Number = 50;
var rect1:Rectangle = new Rectangle(x1, y1, width1, height1);
trace(rect1) // (x=0, y=0, w=100, h=50)
rect1.right = 60;
rect1.bottom = 20;
trace(rect1); // (x=0, y=0, w=60, h=20)

```

也可以使用 `offset()` 方法重新定位 `Rectangle` 对象，如下所示：

```

import flash.geom.Rectangle;
var x1:Number = 0;
var y1:Number = 0;
var width1:Number = 100;
var height1:Number = 50;
var rect1:Rectangle = new Rectangle(x1, y1, width1, height1);
trace(rect1) // (x=0, y=0, w=100, h=50)
rect1.offset(20, 30);
trace(rect1); // (x=20, y=30, w=100, h=50)

```

`offsetPt()` 方法工作方式类似，只不过它是将 `Point` 对象作为参数，而不是将 `x` 和 `y` 偏移量值作为参数。

还可以使用 `inflate()` 方法调整 `Rectangle` 对象的大小，该方法包含两个参数，`dx` 和 `dy`。`dx` 参数表示矩形的左边和右边距中心的像素数，而 `dy` 参数表示矩形的顶边和底边距中心的像素数：

```

import flash.geom.Rectangle;
var x1:Number = 0;
var y1:Number = 0;
var width1:Number = 100;
var height1:Number = 50;
var rect1:Rectangle = new Rectangle(x1, y1, width1, height1);
trace(rect1) // (x=0, y=0, w=100, h=50)
rect1.inflate(6,4);
trace(rect1); // (x=-6, y=-4, w=112, h=58)

```

`inflatePt()` 方法作方式类似，只不过它是将 `Point` 对象作为参数，而不是将 `dx` 和 `dy` 的值作为参数。

确定 `Rectangle` 对象的联合和交集

可以使用 `union()` 方法来确定由两个矩形的边界形成的矩形区域：

```

import flash.display.*;
import flash.geom.Rectangle;
var rect1:Rectangle = new Rectangle(0, 0, 100, 100);
trace(rect1); // (x=0, y=0, w=100, h=100)
var rect2:Rectangle = new Rectangle(120, 60, 100, 100);
trace(rect2); // (x=120, y=60, w=100, h=100)
trace(rect1.union(rect2)); // (x=0, y=0, w=220, h=160)

```

可以使用 `intersection()` 方法来确定由两个矩形重叠区域形成的矩形区域：

```

import flash.display.*;
import flash.geom.Rectangle;
var rect1:Rectangle = new Rectangle(0, 0, 100, 100);
trace(rect1); // (x=0, y=0, w=100, h=100)
var rect2:Rectangle = new Rectangle(80, 60, 100, 100);
trace(rect2); // (x=120, y=60, w=100, h=100)
trace(rect1.intersection(rect2)); // (x=80, y=60, w=20, h=40)

```

使用 `intersects()` 方法查明两个矩形是否相交。也可以使用 `intersects()` 方法查明显示对象是否在舞台的某个区域中。例如，在下面的代码中，假定包含 `circle` 对象的显示对象容器的坐标空间与舞台的坐标空间相同。本示例说明如何使用 `intersects()` 方法来确定显示对象 `circle` 是否与由 `target1` 和 `target2` `Rectangle` 对象定义的指定舞台区域相交：

```

import flash.display.*;
import flash.geom.Rectangle;
var circle:Shape = new Shape();
circle.graphics.lineStyle(2, 0xFF0000);
circle.graphics.drawCircle(250, 250, 100);
addChild(circle);
var circleBounds:Rectangle = circle.getBounds(stage);
var target1:Rectangle = new Rectangle(0, 0, 100, 100);
trace(circleBounds.intersects(target1)); // false
var target2:Rectangle = new Rectangle(0, 0, 300, 300);
trace(circleBounds.intersects(target2)); // true

```

同样，可以使用 `intersects()` 方法查明两个显示对象的边界矩形是否重叠。可以使用 `DisplayObject` 类的 `getRect()` 方法来包括显示对象笔触可添加到边界区域中的其它任何空间。

Rectangle 对象的其它用法

`Rectangle` 对象可用于以下方法和属性：

类	方法或属性	描述
BitmapData	applyFilter()、colorTransform()、copyChannel()、copyPixels()、draw()、fillRect()、generateFilterRect()、getColorBoundsRect()、getPixels()、merge()、paletteMap()、pixelDissolve()、setPixels() 和 threshold()	用作某些参数的类型以定义 <code>BitmapData</code> 对象的区域。
DisplayObject	getBounds()、getRect()、scrollRect、scale9Grid	用作属性的数据类型或返回的数据类型。
PrintJob	addPage()	用于定义 <code>printArea</code> 参数。
Sprite	startDrag()	用于定义 <code>bounds</code> 参数。
TextField	getCharBoundaries()	用作返回值类型。
Transform	pixelBounds	用作数据类型。

使用 Matrix 对象

Matrix 类表示一个转换矩阵，它确定如何将点从一个坐标空间映射到另一个坐标空间。您可以对一个显示对象执行不同的图形转换，方法是设置 Matrix 对象的属性，将该 Matrix 对象应用于 Transform 对象的 matrix 属性，然后应用该 Transform 对象作为显示对象的 transform 属性。这些转换函数包括平移（x 和 y 重新定位）、旋转、缩放和倾斜。

定义 Matrix 对象

虽然可以通过直接调整 Matrix 对象的属性（a、b、c、d、tx 和 ty）来定义矩阵，但更简单的方法是使用 createBox() 方法。使用此方法提供的参数可以直接定义生成的矩阵的缩放、旋转和平移效果。例如，下面的代码创建一个 Matrix 对象，具有效果是水平缩放 2.0、垂直缩放 3.0、旋转 45 度、向右移动（平移）10 个像素并向下移动 20 个像素：

```
var matrix:Matrix = new Matrix();
var scaleX:Number = 2.0;
var scaleY:Number = 3.0;
var rotation:Number = 2 * Math.PI * (45 / 360);
var tx:Number = 10;
var ty:Number = 20;
matrix.createBox(scaleX, scaleY, rotation, tx, ty);
```

还可以使用 scale()、rotate() 和 translate() 方法调整 Matrix 对象的缩放、旋转和平移效果。请注意，这些方法合并了现有 Matrix 对象的值。例如，下面的代码调用两次 scale() 和 rotate() 方法以对 Matrix 对象进行设置，它将对象放大 4 倍并旋转 60 度：

```
var matrix:Matrix = new Matrix();
var rotation:Number = 2 * Math.PI * (30 / 360); // 30°
var scaleFactor:Number = 2;
matrix.scale(scaleFactor, scaleFactor);
matrix.rotate(rotation);
matrix.scale(scaleX, scaleY);
matrix.rotate(rotation);

myDisplayObject.transform.matrix = matrix;
```

要将倾斜转换应用到 Matrix 对象，请调整该对象的 b 或 c 属性。调整 b 属性将矩阵垂直倾斜，并调整 c 属性将矩阵水平倾斜。以下代码使用系数 2 垂直倾斜 myMatrix Matrix 对象：

```
var skewMatrix:Matrix = new Matrix();
skewMatrix.b = Math.tan(2);
myMatrix.concat(skewMatrix);
```

可以将矩阵转换应用到显示对象的 transform 属性。例如，以下代码将矩阵转换应用于名为 myDisplayObject 的显示对象：

```
var matrix:Matrix = myDisplayObject.transform.matrix;
var scaleFactor:Number = 2;
var rotation:Number = 2 * Math.PI * (60 / 360); // 60°
matrix.scale(scaleFactor, scaleFactor);
matrix.rotate(rotation);

myDisplayObject.transform.matrix = matrix;
```

第一行将 Matrix 对象设置为 myDisplayObject 显示对象所使用的现有转换矩阵（myDisplayObject 显示对象的 transformation 属性的 matrix 属性）。这样，调用的 Matrix 类方法将对显示对象的现有位置、缩放和旋转产生累积影响。


注：ColorTransform 类还包含在 flash.geometry 包中。该类用于设置 Transform 对象的 colorTransform 属性。由于它不应有任何种类的几何变形，本章中将不做讨论。有关详细信息，请参阅《ActionScript 3.0 语言和组件参考》中的 ColorTransform 类。

例如：将矩阵转换用于显示对象


DisplayObjectTransformer 范例应用程序说明许多使用 Matrix 类来变换显示对象的功能，包括：

- 旋转显示对象
- 缩放显示对象
- 平移（重新定位）显示对象
- 倾斜显示对象

该应用程序提供了接口，以调整矩阵转换的参数，如下所示：



当用户单击“变形”按钮时，应用程序将应用适当的变形。



原始显示对象和旋转了 -45° 并缩放 50% 的显示对象

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可以在 Samples/DisplayObjectTransformer 中找到 DisplayObjectTransformer 应用程序文件。该应用程序包含以下文件：

文件	描述
DisplayObjectTransformer.mxml 或 DisplayObjectTransformer.fla	Flash (FLA) 或 Flex (MXML) 中的主应用程序文件
com/example/programmingas3/geometry/MatrixTransformer.as	一个类，包含用于应用矩阵转换的方法。
img/	一个目录，包含应用程序使用的范例图像文件。

定义 MatrixTransformer 类

MatrixTransformer 类包含应用 Matrix 对象的几何变形的静态方法。

transform() 方法

transform() 方法包含以下属性的参数：

- sourceMatrix — 此方法转换的输入矩阵
- xScale 和 yScale — x 和 y 缩放系数
- dx 和 dy — x 和 y 平移量（以像素为单位）
- rotation — 旋转量（以度为单位）
- skew — 倾斜系数（以百分比表示）
- skewType — 倾斜的方向，“right”或“left”

返回值为生成的矩阵。

transform() 方法调用下列类的静态方法：

- skew()
- scale()
- translate()
- rotate()

每种方法都返回应用了转换的源矩阵。

skew() 方法

skew() 方法通过调整矩阵的 b 和 c 属性来倾斜矩阵。可选参数 unit 确定用于定义倾斜角度的单位，如果必要，该方法会将 angle 值转换为弧度：

```
if (unit == "degrees")
{
    angle = Math.PI * 2 * angle / 360;
}
if (unit == "gradients")
{
    angle = Math.PI * 2 * angle / 100;
}
```

创建并调整 skewMatrix Matrix 对象以应用倾斜转换。最初，它是恒等矩阵，如下所示：

```
var skewMatrix:Matrix = new Matrix();
```

skewSide 参数确定倾斜应用到的边。如果该参数设置为 “right”，则以下代码设置矩阵的 b 属性：

```
skewMatrix.b = Math.tan(angle);
```

否则，将通过调整矩阵的 c 属性来倾斜底边，如下所示：

```
skewMatrix.c = Math.tan(angle);
```

然后，通过将两个矩阵连接起来以将所产生的倾斜应用到现有矩阵，如下面的示例所示：

```
sourceMatrix.concat(skewMatrix);
return sourceMatrix;
```

scale() 方法

如下面的示例所示，如果提供的缩放系数为百分比，则 scale() 方法将首先调整缩放系数，然后使用矩阵对象的 scale() 方法：

```
if (percent)
{
    xScale = xScale / 100;
    yScale = yScale / 100;
}
sourceMatrix.scale(xScale, yScale);
return sourceMatrix;
```

translate() 方法

translate() 方法只需通过调用矩阵对象的 translate() 方法即可应用 dx 和 dy 平移系数，如下所示：

```
sourceMatrix.translate(dx, dy);
return sourceMatrix;
```

rotate() 方法

rotate() 方法将输入的旋转系数转换为弧度（如果提供的是角度或渐变），然后调用矩阵对象的 rotate() 方法：

```
if (unit == "degrees")
{
    angle = Math.PI * 2 * angle / 360;
}
if (unit == "gradients")
{
    angle = Math.PI * 2 * angle / 100;
}
sourceMatrix.rotate(angle);
return sourceMatrix;
```

从应用程序中调用 MatrixTransformer.transform() 方法

应用程序提供了一个用户界面，以便从用户获得转换参数。然后将这些参数与显示对象的 transform 属性的 matrix 属性一起传递给 Matrix.transform() 方法，如下所示：

```
tempMatrix = MatrixTransformer.transform(tempMatrix,
    xScaleSlider.value,
    yScaleSlider.value,
    dxSlider.value,
    dySlider.value,
    rotationSlider.value,
    skewSlider.value,
    skewSide );
```

然后，该应用程序将返回值应用到显示对象的 transform 属性的 matrix 属性，从而触发转换：

```
img.content.transform.matrix = tempMatrix;
```

第 16 章：过滤显示对象

以往，对位图图像应用滤镜效果是专业图像编辑软件（如 Adobe Photoshop® 和 Adobe Fireworks®）的范畴。

ActionScript 3.0 包括 `flash.filters` 包，它包含一系列位图效果滤镜类，允许开发人员以编程方式对位图应用滤镜并显示对象，以达到图形处理应用程序中所具有的许多相同效果。

过滤显示对象的基础知识

过滤显示对象简介

为应用程序添加优美效果的一种方式是添加简单的图形效果，如在图片后面添加投影可产生三维视觉效果，在按钮周围添加发光可表示该按钮当前处于活动状态。ActionScript 3.0 包括九种可应用于任何显示对象或 `BitmapData` 实例的滤镜。滤镜的范围从基本滤镜（如投影和发光滤镜）到用于创建各种效果的复杂滤镜（如置换图滤镜和卷积滤镜）。

常见过滤任务

以下是您可能会在 ActionScript 中使用滤镜完成的任务：

- 创建滤镜
- 对显示对象应用滤镜
- 删除显示对象中的滤镜
- 对 `BitmapData` 实例中的图像数据应用滤镜
- 删除对象中的滤镜
- 创造各种滤镜效果，如发光、模糊、投影、清晰度、置换、边缘检测、浮雕及其它效果

重要概念和术语

以下参考列表包含将会在本章中遇到的重要术语：

- 斜角：通过使两侧的像素变亮并使相对两侧的像素变暗所形成的一个边缘，它可以产生常用于凸起或凹进按钮和类似图形的三维边界效果。
- 卷积：通过使用各种比率将每个像素的值与其周围某些像素或全部像素的值合并以使图像中的像素发生扭曲。
- 置换：将图像中的像素移动到新位置。
- 矩阵：用于执行特定数学计算的网格数字，使用方法是对网格中的数字应用不同的值，然后合并结果。

完成本章中的示例

学习本章的过程中，您可能想要测试所提供的示例代码清单。由于本章介绍的是创建和操作可视内容，所以测试代码涉及到运行代码以及在创建的 SWF 中查看结果。几乎所有示例都或者使用绘图 API 创建内容，或者加载要应用滤镜的图像。

要测试本章中的代码，请执行以下操作：

- 1 使用 Flash 创作工具创建空白文档
- 2 在时间轴上选择一关键帧。

3 打开“动作”面板，将代码复制到“脚本”窗格中。

4 通过“控制”>“测试影片”运行程序。

您将在所创建的 SWF 文件中看到代码的结果。

几乎所有示例代码都包括用于创建位图图像的代码，因此您可以直接测试代码，而无需提供任何位图内容。或者，您也可以更改代码清单，以加载自己的图像并使用这些图像替代示例中的图像。

创建和应用滤镜

使用滤镜可以对位图和显示对象应用从投影到斜角和模糊等各种效果。由于将每个滤镜定义为一个类，因此应用滤镜涉及创建滤镜对象的实例，这与构造任何其它对象并没有区别。创建了滤镜对象的实例后，通过使用该对象的 `filters` 属性可以很容易地将此实例应用于显示对象；如果是 `BitmapData` 对象，可以使用 `applyFilter()` 方法。

创建新滤镜

若要创建新滤镜对象，只需调用所选的滤镜类的构造函数方法即可。例如，若要创建新的 `DropShadowFilter` 对象，请使用以下代码：

```
import flash.filters.DropShadowFilter;
var myFilter:DropShadowFilter = new DropShadowFilter();
```

虽然此处没有显示参数，但 `DropShadowFilter()` 构造函数（与所有滤镜类的构造函数一样）接受多个可用于自定义滤镜效果外观的可选参数。

应用滤镜

构造滤镜对象后，可以将其应用于显示对象或 `BitmapData` 对象；应用滤镜的方式取决于为之应用该滤镜的对象。

对显示对象应用滤镜

对显示对象应用滤镜效果时，可以通过 `filters` 属性应用这些效果。显示对象的 `filters` 属性是一个 `Array` 实例，其中的元素是应用于该显示对象的滤镜对象。若要对显示对象应用单个滤镜，请创建该滤镜实例，将其添加到 `Array` 实例，再将该 `Array` 对象分配给显示对象的 `filters` 属性：

```
import flash.display.Bitmap;
import flash.display.BitmapData;
import flash.filters.DropShadowFilter;

// Create a bitmapData object and render it to screen
var myBitmapData:BitmapData = new BitmapData(100,100,false,0xFFFF3300);
var myDisplayObject:Bitmap = new Bitmap(myBitmapData);
addChild(myDisplayObject);

// Create a DropShadowFilter instance.
var dropShadow:DropShadowFilter = new DropShadowFilter();

// Create the filters array, adding the filter to the array by passing it as
// a parameter to the Array() constructor.
var filtersArray:Array = new Array(dropShadow);

// Assign the filters array to the display object to apply the filter.
myDisplayObject.filters = filtersArray;
```

如果要为该对象分配多个滤镜，只需在将 **Array** 实例分配给 **filters** 属性之前将所有滤镜添加到该实例即可。可以通过将多个对象作为参数传递给 **Array** 的构造函数，将多个对象添加到 **Array**。例如，以下代码对上面创建的显示对象应用斜角滤镜和发光滤镜：

```
import flash.filters.BevelFilter;
import flash.filters.GlowFilter;

// Create the filters and add them to an array.
var bevel:BevelFilter = new BevelFilter();
var glow:GlowFilter = new GlowFilter();
var filtersArray:Array = new Array(bevel, glow);

// Assign the filters array to the display object to apply the filter.
myDisplayObject.filters = filtersArray;
```

在创建包含滤镜的数组时，您可以使用 **new Array()** 构造函数创建该数组（如前面示例所示），也可以使用 **Array** 文本语法将滤镜括在方括号 (**[]**) 中。例如，下面这行代码：

```
var filters:Array = new Array(dropShadow, blur);
```

与下面这行代码效果相同：

```
var filters:Array = [dropShadow, blur];
```

如果对显示对象应用多个滤镜，则会按顺序以累积方式应用这些滤镜。例如，如果滤镜数组有两个元素：先添加的斜角滤镜和后添加的投影滤镜，则投影滤镜既会应用于斜角滤镜，也会应用于显示对象。这是由于投影滤镜在滤镜数组中位于第二个位置。如果想要以非累积方式应用滤镜，则必须对显示对象的新副本应用每个滤镜。

如果只为显示对象分配一个或几个滤镜，则可以先创建该滤镜实例，然后在单个语句中将该实例分配给该对象。例如，下面的一行代码将模糊滤镜应用于名为 **myDisplayObject** 的显示对象：

```
myDisplayObject.filters = [new BlurFilter()];
```

上面的代码使用 **Array** 文本语法（方括号）创建一个 **Array** 实例，并创建一个新的 **BlurFilter** 实例作为 **Array** 中的一个元素，然后将该 **Array** 分配给名为 **myDisplayObject** 的显示对象的 **filters** 属性。

删除显示对象中的滤镜

删除显示对象中的所有滤镜非常简单，只需为 **filters** 属性分配一个 **null** 值即可：

```
myDisplayObject.filters = null;
```

如果您已对一个对象应用了多个滤镜，并且只想删除其中一个滤镜，则必须完成多个步骤才能更改 **filters** 属性数组。有关详细信息，请参阅第 296 页的“[使用滤镜的潜在问题](#)”。

对 **BitmapData** 对象应用滤镜

对 **BitmapData** 对象应用滤镜需要使用 **BitmapData** 对象的 **applyFilter()** 方法：

```
var rect:Rectangle = new Rectangle();
var origin:Point = new Point();
myBitmapData.applyFilter(sourceBitmapData, rect, origin, new BlurFilter());
```

applyFilter() 方法会对源 **BitmapData** 对象应用滤镜，从而生成一个新的、应用滤镜的图像。此方法不会修改原始的源图像；而是将对源图像应用滤镜的结果存储在调用 **applyFilter()** 方法的 **BitmapData** 实例中。

滤镜的工作原理

显示对象过滤是通过将原始对象的副本缓存为透明位图来工作的。

将滤镜应用于显示对象后，只要此对象具有有效的滤镜列表，**Adobe Flash Player** 或 **Adobe® AIR™** 就会将该对象缓存为位图。然后，将此位图用作所有后续应用的滤镜效果的原始图像。

每个显示对象通常包含两个位图：一个包含原始未过滤的源显示对象，另一个用于过滤后的最终图像。呈现时使用最终图像。只要显示对象不发生更改，最终图像就不需要更新。

使用滤镜的潜在问题

使用滤镜时要记住几个导致混淆或问题的潜在根源。在下面几节中介绍了这些问题。

滤镜和位图缓存

若要对显示对象应用滤镜，必须启用该对象的位图缓存。在对其 `cacheAsBitmap` 属性设置为 `false` 的显示对象应用滤镜时，Flash Player 或 AIR 会自动将该对象的 `cacheAsBitmap` 属性的值设置为 `true`。如果您以后删除了该显示对象中的所有滤镜，Flash Player 或 AIR 会将 `cacheAsBitmap` 属性重置为最后设置的值。

在运行时更改滤镜

如果显示对象已应用了一个或多个滤镜，则不能通过在 `filters` 属性数组中添加或删除滤镜来更改滤镜组。若要添加或更改要应用的滤镜组，必须对单独的数组进行更改，然后将该数组分配给显示对象的 `filters` 属性以对该对象应用滤镜。执行此操作最简单方法是将 `filters` 属性数组读入 `Array` 变量，然后对此临时数组进行修改。然后，将此数组重新分配回显示对象的 `filters` 属性。在较复杂的情况下，可能需要保留一个单独的主滤镜数组。可以对该主滤镜数组进行任何更改，并在每次更改后将该主数组重新分配给显示对象的 `filters` 属性。

添加其它滤镜

下面的代码演示向已应用一个或多个滤镜的显示对象添加其它滤镜的过程。首先，对名为 `myDisplayObject` 的显示对象应用发光滤镜，然后，在单击该显示对象时，调用 `addFilters()` 函数。在此函数中，另有两个滤镜应用于 `myDisplayObject`：

```
import flash.events.MouseEvent;
import flash.filters.*;

myDisplayObject.filters = [new GlowFilter()];

function addFilters(event:MouseEvent):void
{
    // Make a copy of the filters array.
    var filtersCopy:Array = myDisplayObject.filters;

    // Make desired changes to the filters (in this case, adding filters).
    filtersCopy.push(new BlurFilter());
    filtersCopy.push(new DropShadowFilter());

    // Apply the changes by reassigning the array to the filters property.
    myDisplayObject.filters = filtersCopy;
}

myDisplayObject.addEventListener(MouseEvent.CLICK, addFilters);
```

从一组滤镜中删除一个滤镜

如果显示对象已经应用了多个滤镜，您希望删除其中的一个滤镜并希望其它滤镜继续应用于该对象，请将这些滤镜复制到一个临时数组中，从该数组中删除不需要的滤镜，然后将该临时数组重新分配给该显示对象的 `filters` 属性。第 133 页的“[检索值和删除数组元素](#)”一节介绍了从任意数组删除一个或多个元素的多种方法。

最简单的情况是删除对象上最顶层的滤镜（最后一个应用于对象的滤镜）。可以使用 `Array` 类的 `pop()` 方法从数组中删除滤镜：

```
// Example of removing the top-most filter from a display object
// named "filteredObject".

var tempFilters:Array = filteredObject.filters;

// Remove the last element from the Array (the top-most filter).
tempFilters.pop();

// Apply the new set of filters to the display object.
filteredObject.filters = tempFilters;
```

同样，若要删除最底层的滤镜（第一个应用于对象的滤镜），也可以使用相同的代码，不过需要用 `Array` 类的 `shift()` 方法替换 `pop()` 方法。

若要从滤镜数组的中间删除滤镜（假定该数组有两个以上的滤镜），则可以使用 `splice()` 方法。您必须知道要删除的滤镜的索引（在数组中的位置）。例如，下面的代码从显示对象中删除第二个滤镜（索引 1 处的滤镜）：

```
// Example of removing a filter from the middle of a stack of filters
// applied to a display object named "filteredObject".

var tempFilters:Array = filteredObject.filters;

// Remove the second filter from the array. It's the item at index 1
// because Array indexes start from 0.
// The first "1" indicates the index of the filter to remove; the
// second "1" indicates how many elements to remove.
tempFilters.splice(1, 1);

// Apply the new set of filters to the display object.
filteredObject.filters = tempFilters;
```

确定滤镜索引

您需要知道要从数组中删除哪个滤镜，才能确定该滤镜的索引。您必须知道（根据应用程序的设计方式）或计算出要删除的滤镜的索引。

最佳方法是通过设计应用程序，确保要删除的滤镜始终位于滤镜组中的同一位置。例如，如果有一个显示对象先后应用了卷积滤镜和投影滤镜，您希望删除投影滤镜而保留卷积滤镜，由于滤镜位于已知位置（最顶层的滤镜），因此您可以提前知道应使用哪种 `Array` 方法（在本例中，使用 `Array.pop()` 删除投影滤镜）。

如果您要删除的滤镜始终属于某个特定类型，但不一定总在滤镜组中的同一位置，那么您可以检查该数组中每个滤镜的数据类型，从而确定将要删除哪个滤镜。例如，下面的代码确定滤镜组中的哪个滤镜是发光滤镜，然后从该组中删除该滤镜。

```
// Example of removing a glow filter from a set of filters, where the
//filter you want to remove is the only GlowFilter instance applied
// to the filtered object.

var tempFilters:Array = filteredObject.filters;

// Loop through the filters to find the index of the GlowFilter instance.
var glowIndex:int;
var numFilters:int = tempFilters.length;
for (var i:int = 0; i < numFilters; i++)
{
    if (tempFilters[i] is GlowFilter)
    {
        glowIndex = i;
        break;
    }
}

// Remove the glow filter from the array.
tempFilters.splice(glowIndex, 1);

// Apply the new set of filters to the display object.
filteredObject.filters = tempFilters;
```

在更为复杂的情况下（比如要删除的滤镜是在运行时选定的），最好为滤镜数组保留一个单独的永久副本，用作主滤镜列表。不论何时对滤镜组进行更改（添加或删除滤镜），均需更改主列表，然后按照显示对象的 **filters** 属性应用该滤镜数组。

例如，在下面的代码清单中，多个卷积滤镜应用于一个显示对象，以制造不同的视觉效果，随后在应用程序中删除其中一个滤镜而保留其余滤镜。在本例中，代码保留滤镜数组的主副本以及对于要删除的滤镜的引用。查找并删除特定滤镜的方法与前面的方法类似，但不是创建滤镜数组的临时副本，而是操作主副本，然后将其应用于显示对象。

```

// Example of removing a filter from a set of
// filters, where there may be more than one
// of that type of filter applied to the filtered
// object, and you only want to remove one.

// A master list of filters is stored in a separate,
// persistent Array variable.
var masterFilterList:Array;

// At some point, you store a reference to the filter you
// want to remove.
var filterToRemove:ConvolutionFilter;

// ... assume the filters have been added to masterFilterList,
// which is then assigned as the filteredObject.filters:
filteredObject.filters = masterFilterList;

// ... later, when it's time to remove the filter, this code gets called:

// Loop through the filters to find the index of masterFilterList.
var removeIndex:int = -1;
var numFilters:int = masterFilterList.length;
for (var i:int = 0; i < numFilters; i++)
{
    if (masterFilterList[i] == filterToRemove)
    {
        removeIndex = i;
        break;
    }
}

if (removeIndex >= 0)
{
    // Remove the filter from the array.
    masterFilterList.splice(removeIndex, 1);

    // Apply the new set of filters to the display object.
    filteredObject.filters = masterFilterList;
}

```

在此方法（将存储的滤镜引用与滤镜数组中的项目进行比较以确定要删除哪个滤镜）中，必须为滤镜数组保留一个单独的副本 — 如果将存储的滤镜引用与从显示对象的 filters 属性复制的临时数组中的元素进行比较，则代码将不起作用。这是因为在内部将数组分配给 filters 属性时，Flash Player 或 AIR 会为数组中的每个滤镜对象创建副本。这些副本（而非原始对象）将应用于显示对象，而在将 filters 属性读入临时数组时，临时数组包含对复制的滤镜对象的引用，而不是对原始滤镜对象的引用。因此，如果在上一示例中尝试通过将 filterToRemove 与临时滤镜数组中的滤镜进行比较来确定其索引，则不会找到匹配的结果。

滤镜和对象变形

显示对象的边框矩形之外的任何过滤区域（例如，投影）都不能视为可进行点击检测（确定实例是否与其它实例重叠或交叉）的表面。由于 **DisplayObject** 类的点击检测方法是基于矢量的，因此无法对位图结果执行点击检测。例如，如果您对按钮实例应用斜角滤镜，则在该实例的斜角部分，点击检测不可用。

滤镜不支持缩放、旋转和倾斜；如果过滤的显示对象本身进行了缩放（如果 **scaleX** 和 **scaleY** 不是 100%），则滤镜效果将不随该实例缩放。这意味着，实例的原始形状将旋转、缩放或倾斜；而滤镜不随实例一起旋转、缩放或倾斜。

可以使用滤镜给实例添加动画，以形成理想的效果，或者嵌套实例并使用 **BitmapData** 类使滤镜动起来，以获得此效果。

滤镜和位图对象

对 **BitmapData** 对象应用滤镜时，**cacheAsBitmap** 属性会自动设置为 **true**。通过这种方式，滤镜实际上是应用于对象的副本而不是原始对象。

之后，会将此副本放在主显示（原始对象）上，尽量接近最近的像素。如果原始位图的边框发生更改，则会从头重新创建过滤的副本位图，而不是被伸展或扭曲。

如果清除了显示对象的所有滤镜，`cacheAsBitmap` 属性会重置为应用滤镜之前的值。

可用的显示滤镜

ActionScript 3.0 提供十个可用于显示对象和 `BitmapData` 对象的滤镜类：

- 斜角滤镜 (`BevelFilter` 类)
- 模糊滤镜 (`BlurFilter` 类)
- 投影滤镜 (`DropShadowFilter` 类)
- 发光滤镜 (`GlowFilter` 类)
- 渐变斜角滤镜 (`GradientBevelFilter` 类)
- 渐变发光滤镜 (`GradientGlowFilter` 类)
- 颜色矩阵滤镜 (`ColorMatrixFilter` 类)
- 卷积滤镜 (`ConvolutionFilter` 类)
- 置换图滤镜 (`DisplacementMapFilter` 类)
- 着色器滤镜 (`ShaderFilter` 类)

前六个滤镜是简单滤镜，可用于创建一种特定效果，并可以对效果进行某种程度的自定义。可以使用 ActionScript 应用这六个滤镜，也可以在 Adobe Flash CS4 Professional 中使用“滤镜”面板将其应用于对象。因此，即使您要使用 ActionScript 应用滤镜，如果有 Flash 创作工具，也可以使用可视界面快速尝试不同的滤镜和设置，以弄清楚如何创建所需的效果。

最后四个滤镜仅在 ActionScript 中可用。这些滤镜（颜色矩阵滤镜、卷积滤镜、置换图滤镜和着色器滤镜）能够制造的效果类型十分灵活。这些滤镜不是针对一种效果进行优化，而是具有强大的功能和灵活性。例如，如果为卷积滤镜的矩阵选择不同的值，则它可用于创建模糊、浮雕、锐化、查找颜色边缘、变形等效果。

不管是简单滤镜还是复杂滤镜，每个滤镜都可以使用其属性进行自定义。通常，您有两种方法用于设置滤镜属性。所有滤镜都允许通过向滤镜对象的构造函数传递参数值来设置属性。或者，不管您是否通过传递参数来设置滤镜属性，都可以在以后通过设置滤镜对象的属性值来调整滤镜。多数示例代码清单都直接设置属性，以便更易于按照示例进行操作。不过，您通常可以通过在滤镜对象的构造函数中以参数的形式传递值来获得同样的结果。有关每个滤镜及其属性和构造函数参数的详细信息，请参阅 [ActionScript 3.0 语言和组件参考](#) 中的 `flash.filters` 包的列表。

斜角滤镜

使用 `BevelFilter` 类可以为应用了滤镜的对象添加 3D 斜角边缘。此滤镜可使对象的硬角或边缘具有硬角或边缘被凿削或呈斜面的效果。

`BevelFilter` 类属性允许您自定义斜角的外观。您可以设置加亮和阴影颜色、斜角边缘模糊、斜角角度和斜角边缘的位置，甚至可以创建挖空效果。

以下示例加载外部图像并对它应用斜角滤镜。

```
import flash.display.*;
import flash.filters.BevelFilter;
import flash.filters.BitmapFilterQuality;
import flash.filters.BitmapFilterType;
import flash.net.URLRequest;

// Load an image onto the Stage.
var imageLoader:Loader = new Loader();
var url:String = "http://www.helpexamples.com/flash/images/image3.jpg";
var urlReq:URLRequest = new URLRequest(url);
imageLoader.load(urlReq);
addChild(imageLoader);

// Create the bevel filter and set filter properties.
var bevel:BevelFilter = new BevelFilter();

bevel.distance = 5;
bevel.angle = 45;
bevel.highlightColor = 0xFFFF00;
bevel.highlightAlpha = 0.8;
bevel.shadowColor = 0x666666;
bevel.shadowAlpha = 0.8;
bevel.blurX = 5;
bevel.blurY = 5;
bevel.strength = 5;
bevel.quality = BitmapFilterQuality.HIGH;
bevel.type = BitmapFilterType.INNER;
bevel.knockout = false;

// Apply filter to the image.
imageLoader.filters = [bevel];
```

模糊滤镜

BlurFilter 类可使显示对象及其内容具有涂抹或模糊的效果。模糊效果可以用于产生对象不在焦点之内的视觉效果，也可以用于模拟快速运动，比如运动模糊。通过将模糊滤镜的 **quality** 属性设置为低，可以模拟轻轻离开焦点的镜头效果。将 **quality** 属性设置为高会产生类似高斯模糊的平滑模糊效果。

以下示例使用 **Graphics** 类的 **drawCircle()** 方法创建一个圆形对象并对它应用模糊滤镜：

```
import flash.display.Sprite;
import flash.filters.BitmapFilterQuality;
import flash.filters.BlurFilter;

// Draw a circle.
var redDotCutout:Sprite = new Sprite();
redDotCutout.graphics.lineStyle();
redDotCutout.graphics.beginFill(0xFF0000);
redDotCutout.graphics.drawCircle(145, 90, 25);
redDotCutout.graphics.endFill();

// Add the circle to the display list.
addChild(redDotCutout);

// Apply the blur filter to the rectangle.
var blur:BlurFilter = new BlurFilter();
blur.blurX = 10;
blur.blurY = 10;
blur.quality = BitmapFilterQuality.MEDIUM;
redDotCutout.filters = [blur];
```

投影滤镜

投影给人一种目标对象上方有独立光源的印象。可以修改此光源的位置和强度，以产生各种不同的投影效果。

[DropShadowFilter 类](#)所使用的算法与模糊滤镜的算法类似。主要区别是投影滤镜有更多的属性，您可以修改这些属性来模拟不同的光源属性（如 Alpha、颜色、偏移和亮度）。

投影滤镜还允许您对投影的样式应用自定义变形选项，包括内侧或外侧阴影和挖空（也称为剪切块）模式。

以下代码创建方框 `sprite` 并对它应用投影滤镜：

```
import flash.display.Sprite;
import flash.filters.DropShadowFilter;

// Draw a box.
var boxShadow:Sprite = new Sprite();
boxShadow.graphics.lineStyle(1);
boxShadow.graphics.beginFill(0xFF3300);
boxShadow.graphics.drawRect(0, 0, 100, 100);
boxShadow.graphics.endFill();
addChild(boxShadow);

// Apply the drop shadow filter to the box.
var shadow:DropShadowFilter = new DropShadowFilter();
shadow.distance = 10;
shadow.angle = 25;

// You can also set other properties, such as the shadow color,
// alpha, amount of blur, strength, quality, and options for
// inner shadows and knockout effects.

boxShadow.filters = [shadow];
```

发光滤镜

[GlowFilter 类](#)对显示对象应用加亮效果，使显示对象看起来像是被下方的灯光照亮，可创造出一种柔和发光效果。

与投影滤镜类似，发光滤镜包括的属性可修改光源的距离、角度和颜色，以产生各种不同效果。[GlowFilter](#) 还有多个选项用于修改发光样式，包括内侧或外侧发光和挖空模式。

以下代码使用 `Sprite` 类创建一个交叉对象并对它应用发光滤镜：

```
import flash.display.Sprite;
import flash.filters.BitmapFilterQuality;
import flash.filters.GlowFilter;

// Create a cross graphic.
var crossGraphic:Sprite = new Sprite();
crossGraphic.graphics.lineStyle();
crossGraphic.graphics.beginFill(0xCCCC00);
crossGraphic.graphics.drawRect(60, 90, 100, 20);
crossGraphic.graphics.drawRect(100, 50, 20, 100);
crossGraphic.graphics.endFill();
addChild(crossGraphic);

// Apply the glow filter to the cross shape.
var glow:GlowFilter = new GlowFilter();
glow.color = 0x009922;
glow.alpha = 1;
glow.blurX = 25;
glow.blurY = 25;
glow.quality = BitmapFilterQuality.MEDIUM;

crossGraphic.filters = [glow];
```

渐变斜角滤镜

GradientBevelFilter 类可对显示对象或 **BitmapData** 对象应用增强的斜角效果。在斜角上使用渐变颜色可以大大改善斜角的空间深度，使边缘产生一种更逼真的三维外观效果。

以下代码使用 **Shape** 类的 **drawRect()** 方法创建一个矩形对象，并对它应用渐变斜角滤镜。

```
import flash.display.Shape;
import flash.filters.BitmapFilterQuality;
import flash.filters.GradientBevelFilter;

// Draw a rectangle.
var box:Shape = new Shape();
box.graphics.lineStyle();
box.graphics.beginFill(0xFEFE78);
box.graphics.drawRect(100, 50, 90, 200);
box.graphics.endFill();

// Apply a gradient bevel to the rectangle.
var gradientBevel:GradientBevelFilter = new GradientBevelFilter();

gradientBevel.distance = 8;
gradientBevel.angle = 225; // opposite of 45 degrees
gradientBevel.colors = [0xFFFFCC, 0xFEFE78, 0x8F8E01];
gradientBevel.alphas = [1, 0, 1];
gradientBevel.ratios = [0, 128, 255];
gradientBevel.blurX = 8;
gradientBevel.blurY = 8;
gradientBevel.quality = BitmapFilterQuality.HIGH;

// Other properties let you set the filter strength and set options
// for inner bevel and knockout effects.

box.filters = [gradientBevel];

// Add the graphic to the display list.
addChild(box);
```

渐变发光滤镜

GradientGlowFilter 类可对显示对象或 **BitmapData** 对象应用增强的发光效果。该效果可使您更好地控制发光颜色，因而可产生一种更逼真的发光效果。另外，渐变发光滤镜还允许您对对象的内侧、外侧或上侧边缘应用渐变发光。

以下示例在舞台上绘制一个圆形，并对它应用渐变发光滤镜。当您进一步向右和向下移动鼠标时，会分别增加水平和垂直方向的模糊量。此外，只要您在舞台上单击，就会增加模糊的强度。

```
import flash.events.MouseEvent;
import flash.filters.BitmapFilterQuality;
import flash.filters.BitmapFilterType;
import flash.filters.GradientGlowFilter;

// Create a new Shape instance.
var shape:Shape = new Shape();

// Draw the shape.
shape.graphics.beginFill(0xFF0000, 100);
shape.graphics.moveTo(0, 0);
shape.graphics.lineTo(100, 0);
shape.graphics.lineTo(100, 100);
shape.graphics.lineTo(0, 100);
shape.graphics.lineTo(0, 0);
shape.graphics.endFill();

// Position the shape on the Stage.
addChild(shape);
shape.x = 100;
shape.y = 100;

// Define a gradient glow.
var gradientGlow:GradientGlowFilter = new GradientGlowFilter();
gradientGlow.distance = 0;
gradientGlow.angle = 45;
gradientGlow.colors = [0x000000, 0xFF0000];
gradientGlow.alphas = [0, 1];
gradientGlow.ratios = [0, 255];
gradientGlow.blurX = 10;
gradientGlow.blurY = 10;
gradientGlow.strength = 2;
gradientGlow.quality = BitmapFilterQuality.HIGH;
gradientGlow.type = BitmapFilterType.OUTER;

// Define functions to listen for two events.
function onClick(event:MouseEvent):void
{
    gradientGlow.strength++;
    shape.filters = [gradientGlow];
}

function onMouseMove(event:MouseEvent):void
{
    gradientGlow.blurX = (stage.mouseX / stage.stageWidth) * 255;
    gradientGlow.blurY = (stage.mouseY / stage.stageHeight) * 255;
    shape.filters = [gradientGlow];
}

stage.addEventListener(MouseEvent.CLICK, onClick);
stage.addEventListener(MouseEvent.MOUSE_MOVE, onMouseMove);
```

示例：合并基本滤镜

以下代码示例使用与 Timer 合并的多个基本滤镜创建重复动作，以形成一种动画交通信号灯模拟效果。

```
import flash.display.Shape;
import flash.events.TimerEvent;
import flash.filters.BitmapFilterQuality;
import flash.filters.BitmapFilterType;
import flash.filters.DropShadowFilter;
import flash.filters.GlowFilter;
import flash.filters.GradientBevelFilter;
import flash.utils.Timer;

var count:Number = 1;
var distance:Number = 8;
var angleInDegrees:Number = 225; // opposite of 45 degrees
var colors:Array = [0xFFFFCC, 0xFEFE78, 0x8F8E01];
var alphas:Array = [1, 0, 1];
var ratios:Array = [0, 128, 255];
var blurX:Number = 8;
var blurY:Number = 8;
var strength:Number = 1;
var quality:Number = BitmapFilterQuality.HIGH;
var type:String = BitmapFilterType.INNER;
var knockout:Boolean = false;

// Draw the rectangle background for the traffic light.
var box:Shape = new Shape();
box.graphics.lineStyle();
box.graphics.beginFill(0xFEFE78);
box.graphics.drawRect(100, 50, 90, 200);
box.graphics.endFill();

// Draw the 3 circles for the three lights.
var stopLight:Shape = new Shape();
stopLight.graphics.lineStyle();
stopLight.graphics.beginFill(0xFF0000);
stopLight.graphics.drawCircle(145, 90, 25);
stopLight.graphics.endFill();

var cautionLight:Shape = new Shape();
cautionLight.graphics.lineStyle();
cautionLight.graphics.beginFill(0xFF9900);
cautionLight.graphics.drawCircle(145, 150, 25);
cautionLight.graphics.endFill();

var goLight:Shape = new Shape();
goLight.graphics.lineStyle();
goLight.graphics.beginFill(0x00CC00);
goLight.graphics.drawCircle(145, 210, 25);
goLight.graphics.endFill();

// Add the graphics to the display list.
addChild(box);
addChild(stopLight);
addChild(cautionLight);
addChild(goLight);

// Apply a gradient bevel to the traffic light rectangle.
var gradientBevel:GradientBevelFilter = new GradientBevelFilter(distance, angleInDegrees, colors, alphas,
ratios, blurX, blurY, strength, quality, type, knockout);
box.filters = [gradientBevel];

// Create the inner shadow (for lights when off) and glow
// (for lights when on).
var innerShadow:DropShadowFilter = new DropShadowFilter(5, 45, 0, 0.5, 3, 3, 1, 1, true, false);
```

```
var redGlow:GlowFilter = new GlowFilter(0xFF0000, 1, 30, 30, 1, 1, false, false);
var yellowGlow:GlowFilter = new GlowFilter(0xFF9900, 1, 30, 30, 1, 1, false, false);
var greenGlow:GlowFilter = new GlowFilter(0x00CC00, 1, 30, 30, 1, 1, false, false);

// Set the starting state of the lights (green on, red/yellow off).
stopLight.filters = [innerShadow];
cautionLight.filters = [innerShadow];
goLight.filters = [greenGlow];

// Swap the filters based on the count value.
function trafficControl(event:TimerEvent):void
{
    if (count == 4)
    {
        count = 1;
    }

    switch (count)
    {
        case 1:
            stopLight.filters = [innerShadow];
            cautionLight.filters = [yellowGlow];
            goLight.filters = [innerShadow];
            break;
        case 2:
            stopLight.filters = [redGlow];
            cautionLight.filters = [innerShadow];
            goLight.filters = [innerShadow];
            break;
        case 3:
            stopLight.filters = [innerShadow];
            cautionLight.filters = [innerShadow];
            goLight.filters = [greenGlow];
            break;
    }

    count++;
}

// Create a timer to swap the filters at a 3 second interval.
var timer:Timer = new Timer(3000, 9);
timer.addEventListener(TimerEvent.TIMER, trafficControl);
timer.start();
```

颜色矩阵滤镜

[ColorMatrixFilter 类](#)用于操作已应用滤镜的对象的颜色和 Alpha 值。它允许您进行饱和度更改、色相旋转（将调色板从一个颜色范围移动到另一个颜色范围）、将亮度更改为 Alpha，以及生成其它颜色操作效果，方法是使用一个颜色通道中的值，并将这些值潜移默化地应用于其它通道。

从概念上来说，滤镜将逐一处理源图像中的像素，并将每个像素分为红、绿、蓝和 Alpha 组件。然后，用每个值乘以颜色矩阵中提供的值，将结果加在一起以确定该像素将显示在屏幕上的最终颜色值。滤镜的 matrix 属性是一个由 20 个数字组成的数组，用于计算最终颜色。有关用于计算颜色值的特定算法的详细信息，请参阅 [ActionScript 3.0 语言和组件参考](#) 中介绍

[ColorMatrixFilter 类的 matrix 属性](#)条目。

有关颜色矩阵滤镜的其它信息和示例，请参阅 Adobe 开发人员中心网站上提供的文章 “[Using Matrices for Transformations, Color Adjustments, and Convolution Effects in Flash](#)”（在 Flash 中使用矩阵实现变形、颜色调整和卷积效果）。

卷积滤镜

[ConvolutionFilter](#) 类可用于对 `BitmapData` 对象或显示对象应用广泛的图像变形，如模糊、边缘检测、锐化、浮雕和斜角。

从概念上来说，卷积滤镜会逐一处理源图像中的每个像素，并使用像素和它周围的像素的值来确定该像素的最终颜色。指定为数值数组的矩阵可以指示每个特定邻近像素的值对最终结果值具有何种程度的影响。

最常用的矩阵类型是 3×3 矩阵。此矩阵包括九个值：

```
N   N   N  
N   P   N  
N   N   N
```

Flash Player 或 AIR 对特定像素应用卷积滤镜时，会检查该像素本身的颜色值（本示例中的“P”）以及周围像素的值（本示例中的“N”）。而通过设置矩阵中的值，可以指定特定像素在影响生成的图像方面所具有的优先级。

例如，使用卷积滤镜时应用的以下矩阵，会保持图像原样：

```
0   0   0  
0   1   0  
0   0   0
```

图像保持不变的原因是，在决定最终像素颜色时，原始像素的值相对强度为 1，而周围像素的值相对强度为 0（意味着它们的颜色不影响最终图像）。

同样，下面的这个矩阵会使图像的像素向左移动一个像素：

```
0   0   0  
0   0   1  
0   0   0
```

请注意，在本例中，像素本身不影响最终图像上显示在该位置的像素的最终值，而只使用右侧的像素值来确定像素的结果值。

在 ActionScript 中，您可以通过组合一个包含值的 `Array` 实例和两个指定矩阵中行数和列数的属性来创建矩阵。以下示例加载一个图像，完成加载该图像后，使用前面列表中的矩阵对该图像应用卷积滤镜：

```
// Load an image onto the Stage.  
var loader:Loader = new Loader();  
var url:URLRequest = new URLRequest("http://www.helpexamples.com/flash/images/image1.jpg");  
loader.load(url);  
this.addChild(loader);  
  
function applyFilter(event:MouseEvent):void  
{  
    // Create the convolution matrix.  
    var matrix:Array = [0, 0, 0,  
                      0, 0, 1,  
                      0, 0, 0];  
  
    var convolution:ConvolutionFilter = new ConvolutionFilter();  
    convolution.matrixX = 3;  
    convolution.matrixY = 3;  
    convolution.matrix = matrix;  
    convolution.divisor = 1;  
  
    loader.filters = [convolution];  
}  
  
loader.addEventListener(MouseEvent.CLICK, applyFilter);
```

此代码中没有明确显示使用除矩阵中 1 或 0 以外的值将会产生的效果。例如，同一矩阵中如果右侧位置的数字是 8 而不是 1，则它会执行同样的动作（向左移动像素）。此外，它还影响图像的颜色，使颜色加亮了 8 倍。这是因为最终像素颜色值的计算方法是：用原始像素颜色乘以矩阵值，将这些值加在一起，再除以滤镜的 divisor 属性的值。请注意，在示例代码中， divisor 属性设置为 1。通常，如果想让颜色的明亮度与原始图像保持基本相同，应让 divisor 等于矩阵值之和。因此，如果矩阵的值相加为 8，并且除数为 1，结果图像将比原始图像明亮约 8 倍。

虽然此矩阵的效果不是很明显，但可以使用其它矩阵值以产生各种不同效果。以下是几组标准矩阵值集合，用于使用 3 x 3 矩阵产生不同效果：

- 基本模糊（除数 5）：

```
0 1 0
1 1 1
0 1 0
```

- 锐化（除数 1）：

```
0, -1, 0
-1, 5, -1
0, -1, 0
```

- 边缘检测（除数 1）：

```
0, -1, 0
-1, 4, -1
0, -1, 0
```

- 浮雕效果（除数 1）：

```
-2, -1, 0
-1, 1, 1
0, 1, 2
```

请注意，对于上述大部分效果， divisor 均为 1。这是因为负矩阵值加正矩阵值产生 1（或边缘检测中的 0，但 divisor 属性的值不能为 0）。

置换图滤镜

[DisplacementMapFilter](#) 类使用 `BitmapData` 对象（称为置换图图像）的像素值在新对象上执行置换效果。通常，置换图图像与将要应用滤镜的实际显示对象或 `BitmapData` 实例不同。置换效果包括置换经过过滤的图像中的像素，即让这些像素离开各自原始位置一定距离。此滤镜可用于产生移位、扭曲或斑点效果。

应用于给定像素的置换位置和置换量由置换图图像的颜色值确定。使用滤镜时，除了指定置换图图像外，还要指定以下值，以便控制置换图图像中计算置换的方式：

- 映射点：过滤图像上的位置，在该点将应用置换滤镜的左上角。如果只想对图像的一部分应用滤镜，可以使用此值。
- X 组件：影响像素的 x 位置的置换图图像的颜色通道。
- Y 组件：影响像素的 y 位置的置换图图像的颜色通道。
- X 缩放比例：指定 x 轴置换强度的乘数值。
- Y 缩放比例：指定 y 轴置换强度的乘数值。
- 滤镜模式：确定在移开像素后形成的空白区域中， Flash Player 或 AIR 应执行什么操作。在 `DisplacementMapFilterMode` 类中定义为常量的选项可以显示原始像素（滤镜模式 IGNORE）、从图像的另一侧环绕像素（滤镜模式 WRAP，这是默认设置）、使用最近的移位像素（滤镜模式 CLAMP）或用颜色填充空间（滤镜模式 COLOR）。

若要了解置换图滤镜的工作原理，请查看下面的基本示例。在以下代码中，将加载一个图像，完成加载后使图像在舞台上居中并对它应用置换图滤镜，使整个图像中的像素向左水平移位。

```
import flash.display.BitmapData;
import flash.display.Loader;
import flash.events.MouseEvent;
import flash.filters.DisplacementMapFilter;
import flash.geom.Point;
import flash.net.URLRequest;

// Load an image onto the Stage.
var loader:Loader = new Loader();
var url:URLRequest = new URLRequest("http://www.helpexamples.com/flash/images/image3.jpg");
loader.load(url);
this.addChild(loader);

var mapImage:BitmapData;
var displacementMap:DisplacementMapFilter;

// This function is called when the image finishes loading.
function setupStage(event:Event):void
{
    // Center the loaded image on the Stage.
    loader.x = (stage.stageWidth - loader.width) / 2;
    loader.y = (stage.stageHeight - loader.height) / 2;

    // Create the displacement map image.
    mapImage = new BitmapData(loader.width, loader.height, false, 0xFF0000);

    // Create the displacement filter.
    displacementMap = new DisplacementMapFilter();
    displacementMap.mapBitmap = mapImage;
    displacementMap.mapPoint = new Point(0, 0);
    displacementMap.componentX = BitmapDataChannel.RED;
    displacementMap.scaleX = 250;
    loader.filters = [displacementMap];
}

loader.contentLoaderInfo.addEventListener(Event.COMPLETE, setupStage);
```

用于定义置换的属性有：

- 置换位图：置换位图是由代码创建的新 `BitmapData` 实例。它的尺寸与加载的图像的尺寸匹配（因此会将置换应用于整个图像）。用纯红色像素填充此实例。
- 映射点：将此值设置为点 `0, 0`，使置换再次应用于整个图像。
- X 组件：此值设置为常量 `BitmapDataChannel.RED`，表示置换图位图的红色值将决定沿着 x 轴置换像素的程度（像素的移动程度）。
- X 缩放比例：此值设置为 `250`。由于全部置换量（与全红的置换图图像的距离）仅使图像置换很小的量（大约为一个像素的一半），因此，如果将此值设置为 `1`，图像只会水平移动 `0.5` 个像素。将此值设置为 `250`，图像将移动大约 `125` 个像素。

这些设置可使过滤图像的像素向左移动 `250` 个像素。移动的方向（向左或向右）和移动量取决于置换图图像中的像素的颜色值。从概念上来说，Flash Player 或 AIR 会逐一处理过滤图像的像素（至少处理将应用滤镜的区域中的像素，在本例中指所有像素），并对每个像素执行以下操作：

- 1 Flash Player 在置换图图像中查找相应的像素。例如，当 Flash Player 或 AIR 计算过滤图像左上角像素的置换量时，会在置换图图像左上角中查找像素。
- 2 Flash Player 确定置换图像素中指定颜色通道的值。在本例中，x 组件颜色通道是红色通道，因此 Flash Player 和 AIR 将查看置换图图像中该像素所在位置处的红色通道所对应的值。由于置换图图像是纯红色的，所以像素的红色通道为 `0xFF`（即 `255`）。该值将用作置换值。

- 3 比较置换值和“中间”值(127,它是0和255之间的中间值)。如果置换值低于中间值,则像素正向移位(x置换向右;y置换向下)。另一方面,如果置换值高于中间值(如本示例),则像素负向移位(x置换向左;y置换向上)。为更精确起见,Flash Player和AIR会从127中减去置换值,结果(正或负)即是应用的相对置换量。
- 4 最后,Flash Player通过确定相对置换值所表示的完全置换量的百分比来确定实际置换量。在本例中,全红色意味着100%置换。然后用x缩放比例值或y缩放比例值乘以该百分比,以确定将应用的置换像素数。在本示例中,100%乘以一个乘数250可以确定置换量(大约为向左移动125个像素)。

因为没有为y分量和y缩放比例指定值,所以使用默认值(不发生置换),这就是图像在垂直方向不移位的原因。

由于在本示例中使用了默认滤镜模式设置WRAP,因此在像素向左移位时,会用移到图像左边缘以外的像素填充右侧空白区域。您可以对此设置试用不同的值以查看不同的效果。例如,如果您在设置置换属性的代码部分中(在loader.filters=[displacementMap]行之前)添加以下一行内容,则会使图像在舞台上产生涂抹的效果:

```
displacementMap.mode = DisplacementMapFilterMode.CLAMP;
```

下面是一个更为复杂的示例,代码清单中使用置换图滤镜在图像上创建放大镜效果:

```
import flash.display.Bitmap;
import flash.display.BitmapData;
import flash.display.BitmapDataChannel;
import flash.display.GradientType;
import flash.display.Loader;
import flash.display.Shape;
import flash.events.MouseEvent;
import flash.filters.DisplacementMapFilter;
import flash.filters.DisplacementMapFilterMode;
import flash.geom.Matrix;
import flash.geom.Point;
import flash.net.URLRequest;

// Create the gradient circles that will together form the
// displacement map image
var radius:uint = 50;

var type:String = GradientType.LINEAR;
var redColors:Array = [0xFF0000, 0x000000];
var blueColors:Array = [0x0000FF, 0x000000];
var alphas:Array = [1, 1];
var ratios:Array = [0, 255];
var xMatrix:Matrix = new Matrix();
xMatrix.createGradientBox(radius * 2, radius * 2);
var yMatrix:Matrix = new Matrix();
yMatrix.createGradientBox(radius * 2, radius * 2, Math.PI / 2);

var xCircle:Shape = new Shape();
xCircle.graphics.lineStyle(0, 0, 0);
xCircle.graphics.beginGradientFill(type, redColors, alphas, ratios, xMatrix);
xCircle.graphics.drawCircle(radius, radius, radius);

var yCircle:Shape = new Shape();
yCircle.graphics.lineStyle(0, 0, 0);
yCircle.graphics.beginGradientFill(type, blueColors, alphas, ratios, yMatrix);
yCircle.graphics.drawCircle(radius, radius, radius);

// Position the circles at the bottom of the screen, for reference.
this.addChild(xCircle);
xCircle.y = stage.stageHeight - xCircle.height;
this.addChild(yCircle);
yCircle.y = stage.stageHeight - yCircle.height;
yCircle.x = 200;
```

```
// Load an image onto the Stage.  
var loader:Loader = new Loader();  
var url:URLRequest = new URLRequest("http://www.helpexamples.com/flash/images/image1.jpg");  
loader.load(url);  
this.addChild(loader);  
  
// Create the map image by combining the two gradient circles.  
var map:BitmapData = new BitmapData(xCircle.width, xCircle.height, false, 0x7F7F7F);  
map.draw(xCircle);  
var yMap:BitmapData = new BitmapData(yCircle.width, yCircle.height, false, 0x7F7F7F);  
yMap.draw(yCircle);  
map.copyChannel(yMap, yMap.rect, new Point(0, 0), BitmapDataChannel.BLUE, BitmapDataChannel.BLUE);  
yMap.dispose();  
  
// Display the map image on the Stage, for reference.  
var mapBitmap:Bitmap = new Bitmap(map);  
this.addChild(mapBitmap);  
mapBitmap.x = 400;  
mapBitmap.y = stage.stageHeight - mapBitmap.height;  
  
// This function creates the displacement map filter at the mouse location.  
function magnify():void  
{  
    // Position the filter.  
    var filterX:Number = (loader.mouseX) - (map.width / 2);  
    var filterY:Number = (loader.mouseY) - (map.height / 2);  
    var pt:Point = new Point(filterX, filterY);  
    var xyFilter:DisplacementMapFilter = new DisplacementMapFilter();  
    xyFilter.mapBitmap = map;  
    xyFilter.mapPoint = pt;  
    // The red in the map image will control x displacement.  
    xyFilter.componentX = BitmapDataChannel.RED;  
    // The blue in the map image will control y displacement.  
    xyFilter.componentY = BitmapDataChannel.BLUE;  
    xyFilter.scaleX = 35;  
    xyFilter.scaleY = 35;  
    xyFilter.mode = DisplacementMapFilterMode.IGNORE;  
    loader.filters = [xyFilter];  
}  
  
// This function is called when the mouse moves. If the mouse is  
// over the loaded image, it applies the filter.  
function moveMagnifier(event:MouseEvent):void  
{  
    if (loader.hitTestPoint(loader.mouseX, loader.mouseY))  
    {  
        magnify();  
    }  
}  
loader.addEventListener(MouseEvent.MOUSE_MOVE, moveMagnifier);
```

代码首先生成两个渐变圆，它们合并在一起构成置换图图像。红色圆创建 x 轴置换 (xyFilter.componentX = BitmapDataChannel.RED)，蓝色圆创建 y 轴置换 (xyFilter.componentY = BitmapDataChannel.BLUE)。为帮助您理解置换图图像的外观，代码向屏幕底部添加了原始圆形以及合并后作为置换图图像的圆形。



然后，代码加载一个图像，当鼠标移动时，将置换滤镜应用于鼠标下方的图像部分。用作置换图图像的渐变圆使置换区域从鼠标指针处向外扩展。请注意，置换图图像的灰色区域不会发生置换。灰色为 0x7F7F7F。该灰色的蓝色和红色通道与这些颜色通道中的中间色调完全匹配，因此在置换图图像的灰色区域不会发生置换。同样，在圆的中心也不会发生置换。由于蓝色和红色是引起置换的颜色，虽然颜色中没有灰色，但该颜色的蓝色通道和红色通道与中度灰的蓝色通道和红色通道完全相同，因此该处不发生置换。

着色器滤镜

[ShaderFilter](#) 类可使用定义为 Pixel Bender 着色器的自定义滤镜效果。由于该滤镜效果是以 Pixel Bender 着色器形式编写的，因此可完全自定义。过滤内容将作为图像输入传递给着色器，着色器操作的结果就是滤镜结果。

注：从 Flash Player 10 和 Adobe AIR 1.5 开始，在 ActionScript 中提供着色器滤镜。

若要对某个对象应用着色器滤镜，首先应创建一个表示要使用的 Pixel Bender 着色器的 [Shader](#) 实例。有关 [Shader](#) 实例创建过程以及如何指定输入图像和参数值的详细信息，请参阅第 319 页的“[使用 Pixel Bender 着色器](#)”。

将着色器用作滤镜时，请记住以下三个要点：

- 必须将着色器定义为至少接受一个输入图像。
- 将过滤对象（为之应用滤镜的显示对象或 [BitmapData](#) 对象）作为第一个输入图像值传递给着色器。因此，不应为第一个图像输入手动指定值。
- 如果着色器定义了多个输入图像，则必须手动指定其它输入（即，为属于 [Shader](#) 实例的任何 [ShaderInput](#) 实例设置 [input](#) 属性）。

创建着色器的 [Shader](#) 对象后，即创建了一个 [ShaderFilter](#) 实例。这就是使用方法与其它所有滤镜相同的实际滤镜对象。若要创建使用 [Shader](#) 对象的 [ShaderFilter](#)，请调用 [ShaderFilter\(\)](#) 构造函数，并将 [Shader](#) 对象作为参数传递，如下所示：

```
var myFilter:ShaderFilter = new ShaderFilter(myShader);
```

有关使用着色器滤镜的完整示例，请参阅第 333 页的“[使用着色器作为滤镜](#)”。

示例：Filter Workbench

Filter Workbench 提供了一个用户界面，用于对图像和其它可视内容应用不同的滤镜以及查看结果代码，这些代码可用于在 ActionScript 中生成相同的效果。除了提供试验滤镜的工具外，此应用程序还展示了以下技巧：

- 创建各种滤镜的实例
- 对显示对象应用多种滤镜

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可以在 Samples/FilterWorkbench 文件夹中找到 Filter Workbench 应用程序文件。该应用程序包含以下文件：

文件	描述
com/example/programmingas3/filterWorkbench/FilterWorkbenchController.as	提供应用程序主要功能的类，主要功能包括切换要应用滤镜的内容以及将滤镜应用于内容。
com/example/programmingas3/filterWorkbench/IFilterFactory.as	用来定义由各滤镜工厂类所实现的常用方法的接口。此接口定义 FilterWorkbenchController 类用来与各个滤镜工厂类交互的常见功能。
在文件夹 com/example/programmingas3/filterWorkbench/ 中： BevelFactory.as BlurFactory.as ColorMatrixFactory.as ConvolutionFactory.as DropShadowFactory.as GlowFactory.as GradientBevelFactory.as GradientGlowFactory.as	类集，每个类都实现 IFilterFactory 接口。每个类均可为某一类型的滤镜提供创建和设置值的功能。应用程序中的滤镜属性面板使用这些工厂类来创建其特定滤镜的实例，FilterWorkbenchController 类将检索这些实例并将它们应用于图像内容。
com/example/programmingas3/filterWorkbench/IFilterPanel.as	定义某些类（这些类定义用于操作应用程序中滤镜值的用户界面面板）所实现的常见方法的接口。
com/example/programmingas3/filterWorkbench/ColorStringFormatter.as	实用程序类，包含将数字颜色值转换为十六进制字符串格式的方法
com/example/programmingas3/filterWorkbench/GradientColor.as	作为值对象的类，此类将与 GradientBevelFilter 和 GradientGlowFilter 中各颜色相关联的三个值（颜色、Alpha 和比例）合并到单个对象中
用户界面 (Flash)	
FilterWorkbench.fla	定义应用程序用户界面的主要文件。
flashapp/FilterWorkbench.as	为主应用程序用户界面提供功能的类；此类用作应用程序 FLA 文件的文档类。


文件	描述
在 flashapp/filterPanels 文件夹中:	
BevelPanel.as	类集，为各面板（用于为单个滤镜设置选项）提供功能。
BlurPanel.as	对于每个类而言，在主应用程序 FLA 文件库中，还有一个关联的 MovieClip 元件，其名称与类的名称相匹配（例如，“BlurPanel”元件链接到 BlurPanel.as 中定义的类）。构成用户界面的组件在这些元件中进行定位和命名。
ColorMatrixPanel.as	
ConvolutionPanel.as	
DropShadowPanel.as	
GlowPanel.as	
GradientBevelPanel.as	
GradientGlowPanel.as	
flashapp/ImageContainer.as	作为屏幕上已加载图像的容器的显示对象
flashapp/BGColorCellRenderer.as	用于更改 DataGrid 组件中单元格的背景颜色的自定义单元格渲染器
flashapp/ButtonCellRenderer.as	用于将 Button 组件包含在 DataGrid 组件的单元格中的自定义单元格渲染器
已过滤的图像内容	
com/example/programmingas3/filterWorkbench/ImageType.as	此类用作值对象，它包含应用程序可加载并对其应用滤镜的单个图像文件的类型和 URL。此类还包含一组代表实际可用图像文件的常数。
images/sampleAnimation.swf	应用程序中应用滤镜的图像和其它可视内容。
images/sampleImage1.jpg	
images/sampleImage2.jpg	

试验 ActionScript 滤镜

Filter Workbench 应用程序旨在帮助您试验各种滤镜效果，并为该效果生成相关的 ActionScript 代码。使用该应用程序，您可以从包含可视内容（包括位图图像和 Flash 创建的动画）的三种不同文件中进行选择，并可采用单独或与其它滤镜相结合的方式，将八种不同的 ActionScript 滤镜应用于选定的图像。此应用程序包含以下滤镜：

- 斜角 (flash.filters.BevelFilter)
- 模糊 (flash.filters.BlurFilter)
- 颜色矩阵 (flash.filters.ColorMatrixFilter)
- 卷积 (flash.filters.ConvolutionFilter)
- 投影 (flash.filters.DropShadowFilter)
- 发光 (flash.filters.GlowFilter)
- 渐变斜角 (flash.filters.GradientBevelFilter)
- 渐变发光 (flash.filters.GradientGlowFilter)

用户选定图像并将滤镜应用于图像后，应用程序将显示一个带有用于设置选定滤镜特定属性的控件的面板。例如，下列图像显示的是已选择“斜角”滤镜的应用程序：



如果用户调整滤镜属性，则预览将实时更新。用户还可以应用多个滤镜，方法是：自定义一个滤镜，单击“应用”按钮；自定义另一个滤镜，单击“应用”按钮，以此类推。

应用程序的滤镜面板具有一些功能和限制：

- 颜色矩阵滤镜包含一组用于直接操作常用图像属性（包括亮度、对比度、饱和度和色相）的控件。另外，可以指定自定义颜色矩阵值。
- 卷积滤镜只有在使用 ActionScript 时才可用，此滤镜包含一组常用的卷积矩阵值，也可指定自定义值。但是，虽然 ConvolutionFilter 类可以接受任意大小的矩阵，但 Filter Workbench 应用程序使用的是固定的 3 x 3 矩阵，这是最常用的滤镜大小。
- 置换图滤镜和着色器滤镜仅在 ActionScript 中可用，在 Filter Workbench 应用程序中不可用。置换图滤镜除要求使用已过滤的图像内容外，还要求使用映射图像。置换图滤镜的置换图像是决定滤镜效果的主要输入，因此，如果不具有加载或创建置换图图像的功能，则对置换图滤镜进行的试验将受到极大限制。同样，除了经过过滤的图像内容，着色器滤镜还需要 Pixel Bender 字节码文件。如果不能加载着色器字节码，则无法使用着色器滤镜。

创建滤镜实例

Filter Workbench 应用程序包含一个类集，各可用滤镜均有一个对应的类，各面板用这些类来创建滤镜。用户选定滤镜后，与滤镜面板相关联的 ActionScript 代码就将创建相应滤镜工厂类的实例。（这些类称为工厂类，因为其用途在于创建其它对象的实例，这与实际工厂创建各产品极为相似）

只要用户更改面板上的属性值，面板的代码就会调用工厂类中的相应方法。各工厂类包含面板用于创建相应滤镜实例的特定方法。例如，如果用户选择“模糊”滤镜，则应用程序将创建 BlurFactory 实例。BlurFactory 类包含一个 modifyFilter() 方法，该方法接受三个参数：blurX、blurY 和 quality，这三个参数一起用于创建所需的 BlurFilter 实例：

```

private var _filter:BlurFilter;

public function modifyFilter(blurX:Number = 4, blurY:Number = 4, quality:int = 1):void
{
    _filter = new BlurFilter(blurX, blurY, quality);
    dispatchEvent(new Event(Event.CHANGE));
}

```

另一方面，如果用户选择“卷积”滤镜，因为该滤镜允许更大的灵活性，所以需要控制的属性就会更多。在 ConvolutionFactory 类中，如果用户在滤镜面板上选择不同的值，则将调用以下代码：

```

private var _filter:ConvolutionFilter;

public function modifyFilter(matrixX:Number = 0,
                            matrixY:Number = 0,
                            matrix:Array = null,
                            divisor:Number = 1.0,
                            bias:Number = 0.0,
                            preserveAlpha:Boolean = true,
                            clamp:Boolean = true,
                            color:uint = 0,
                            alpha:Number = 0.0):void
{
    _filter = new ConvolutionFilter(matrixX, matrixY, matrix, divisor, bias, preserveAlpha, clamp, color,
alpha);
    dispatchEvent(new Event(Event.CHANGE));
}

```

请注意，在各例中，如果滤镜值发生变化，则工厂对象将调度 Event.CHANGE 事件，从而向侦听器通知滤镜的值已经改变。将滤镜实际应用于过滤内容的 FilterWorkbenchController 类将侦听该事件，以确定何时需检索滤镜的新副本并将其重新应用于过滤的内容。

FilterWorkbenchController 类无需知道各滤镜工厂类的特定详细信息，它只需知道滤镜已更改，并能够访问滤镜副本即可。为支持此目的，应用程序包含了一个 IFilterFactory 接口，该接口定义滤镜工厂类需提供的行为，以便应用程序的 FilterWorkbenchController 实例能够完成自己的工作。IFilterFactory 定义在 FilterWorkbenchController 类中使用的 getFilter() 方法：

```
function getFilter():BitmapFilter;
```

请注意，getFilter() 接口方法定义指定该方法返回一个 BitmapFilter 实例，而非特定类型的滤镜。BitmapFilter 类不定义特定类型的滤镜，而是用来构建所有滤镜类的基类。各滤镜工厂类定义 getFilter() 方法的特定实现，此时将返回对所创建滤镜对象的引用。例如，以下就是 ConvolutionFactory 类源代码的缩写版本：

```

public class ConvolutionFactory extends EventDispatcher implements IFilterFactory
{
    // ----- Private vars -----
    private var _filter:ConvolutionFilter;
    ...

    // ----- IFilterFactory implementation -----
    public function getFilter():BitmapFilter
    {
        return _filter;
    }
    ...
}

```

ConvolutionFactory 类在实现 getFilter() 方法时将返回一个 ConvolutionFilter 实例，尽管调用 getFilter() 的任何对象无需知道此返回结果。根据 ConvolutionFactory 所遵循的 getFilter() 方法的定义，此方法必须返回任意 BitmapFilter 实例，这些实例可能是任意 ActionScript 滤镜类的实例。

将滤镜应用于显示对象

正如上一部分所说明的那样，Filter Workbench 应用程序使用 FilterWorkbenchController 类的实例（下文称为“控制器实例”），该实例执行将滤镜应用于所选可视对象的实际任务。控制器实例必须首先知道滤镜需应用于哪些图像或可视内容，然后才能应用滤镜。用户选定图像后，应用程序将调用 FilterWorkbenchController 类中的 setFilterTarget() 方法，传入在 ImageType 类中定义的一个常数：

```
public function setFilterTarget(targetType:ImageType):void
{
    ...
    _loader = new Loader();
    ...
    _loader.contentLoaderInfo.addEventListener(Event.COMPLETE, targetLoadComplete);
    ...
}
```

控制器实例使用该信息来加载指定的文件，并将所加载的文件存储在一个名为 _currentTarget 的实例变量中：

```
private var _currentTarget:DisplayObject;

private function targetLoadComplete(event:Event):void
{
    ...
    _currentTarget = _loader.content;
    ...
}
```

用户选择滤镜后，应用程序将调用控制器实例的 setFilter() 方法，为控制器提供对相关滤镜工厂对象的引用，此引用存储在名为 _filterFactory 的实例变量中。

```
private var _filterFactory:IFilterFactory;

public function setFilter(factory:IFilterFactory):void
{
    ...
    _filterFactory = factory;
    _filterFactory.addEventListener(Event.CHANGE, filterChange);
}
```

请注意，正如上文所说明的那样，控制器实例并不知道为其提供的滤镜工厂实例的具体数据类型，它只知道对象实现 IFilterFactory 实例，这意味着它具有 getFilter() 方法，并可在滤镜更改时调度 change (Event.CHANGE) 事件。

如果用户改变滤镜面板中的滤镜属性，则控制器实例将通过滤镜工厂的 change 事件发现滤镜已发生更改，该事件调用控制器实例的 filterChange() 方法。而该方法又将调用 applyTemporaryFilter() 方法：

```
private function filterChange(event:Event):void
{
    applyTemporaryFilter();
}

private function applyTemporaryFilter():void
{
    var currentFilter:BitmapFilter = _filterFactory.getFilter();

    // Add the current filter to the set temporarily
    _currentFilters.push(currentFilter);

    // Refresh the filter set of the filter target
    _currentTarget.filters = _currentFilters;

    // Remove the current filter from the set
    // (This doesn't remove it from the filter target, since
    // the target uses a copy of the filters array internally.)
    _currentFilters.pop();
}
```

将滤镜应用于显示对象的工作在 `applyTemporaryFilter()` 方法中进行。首先，控制器将通过调用滤镜工厂的 `getFilter()` 方法来检索对滤镜对象的引用。

```
var currentFilter:BitmapFilter = _filterFactory.getFilter();
```

控制器实例有一个名为 `_currentFilters` 的数组实例变量，所有曾应用于显示对象的滤镜均存储在该变量中。下一步就是将新近更新的滤镜添加到该数组：

```
_currentFilters.push(currentFilter);
```

然后，代码将滤镜数组分配给显示对象的 `filters` 属性，这会将滤镜实际应用于图像：

```
_currentTarget.filters = _currentFilters;
```

最后，由于此最近添加的滤镜依旧是“工作”滤镜，不应将其永久应用于显示对象，因此将其从 `_currentFilters` 数组中删除：

```
_currentFilters.pop();
```

将此滤镜从数组中删除并不会影响已过滤的显示对象，因为在将滤镜数组分配给 `filters` 属性时，显示对象将生成滤镜数组的副本，而且显示对象使用的是内部数组而非原始数组。因此，对滤镜数组所做的任何更改均不会影响显示对象，直至数组被再次分配给显示对象的 `filters` 属性。

第 17 章：使用 Pixel Bender 着色器

利用 Adobe Pixel Bender 工具包，开发人员可以编写用于创建图形效果、执行其它图像和数据处理的着色器。Pixel Bender 字节码可以在 ActionScript 中执行，从而将效果应用于图像数据或可视内容。通过在 ActionScript 中使用 Pixel Bender 着色器，不仅可以运用 ActionScript 的内置功能，而且还可以创建自定义的视觉效果并执行数据处理。

注：从 Flash Player 10 和 Adobe AIR 1.5 开始支持 Pixel Bender。

Pixel Bender 着色器基础知识

Pixel Bender 着色器使用简介

Adobe Pixel Bender 是一种编程语言，用于创建或操作图像内容。您可以使用 Pixel Bender 创建一个内核（在本文档中亦称之为着色器）。着色器定义了一个可对图像的每个像素单独执行的单一函数。对该函数的每次调用都将得到图像中该像素坐标处的输出颜色。可通过指定输入图像和参数值来自定义该操作。在着色器的单次执行中，输入值和参数值是不变的。唯一发生变化的是像素（其颜色是函数调用的结果）的坐标。

对多个输出像素坐标调用着色器函数时，将尽可能采用并行方式。这样会改进着色器性能，提供高性能的处理能力。

在 Flash Player 和 Adobe AIR 中，使用着色器可轻松创建三种类型的效果：

- 绘制填充
- 混合模式
- 滤镜

着色器也可以按独立模式执行。使用独立模式时，将直接访问着色器的结果，而非预先指定着色器的用途。结果可以按图像数据或者二进制或数值数据的形式访问。这样一来，您可以为着色器输入一组数据。着色器将处理该数据，然后您可以访问着色器返回的结果数据。

注：从 Flash Player 10 和 Adobe AIR 1.5 开始支持 Pixel Bender。

Pixel Bender 着色器常见任务

以下是您可能会在 ActionScript 中使用滤镜完成的任务：

- 将着色器加载到正在运行的 SWF 应用程序中，或者在编译时嵌入着色器并在运行时访问它。
- 访问着色器元数据
- 为着色器标识并指定输入值（通常为图像）
- 为着色器标识并指定参数值
- 着色器有以下几种使用方法：
 - 用于绘制填充
 - 用于混合模式
 - 用作滤镜
 - 按独立模式使用

重要概念和术语

以下参考列表包含本章中会遇到的重要术语：

- 内核：对于 Pixel Bender 而言，内核指的就是着色器。通过 Pixel Bender，您的代码定义了一个内核，它定义了可对图像的每个像素单独执行的单一函数。
- Pixel Bender 字节代码：编译 Pixel Bender 内核时，会将其转换为 Pixel Bender 字节代码。Flash Player 或 Adobe AIR 在运行时访问并执行字节代码。
- Pixel Bender 语言：用于创建 Pixel Bender 内核的编程语言。
- Pixel Bender 工具包：用于依据 Pixel Bender 源代码创建 Pixel Bender 字节代码文件的应用程序。您可以使用该工具包编写、测试和编译 Pixel Bender 源代码。
- 着色器：在本文档中，着色器是指采用 Pixel Bender 语言编写的一组功能。着色器的代码会创建视觉效果或执行计算。在任一情况下，着色器都返回一组数据（通常为图像的像素）。着色器对每个数据点执行相同的操作，唯一的区别在于输出像素的坐标不同。

着色器不是用 ActionScript 编写的。它用 Pixel Bender 语言编写，并编译为 Pixel Bender 字节代码。着色器可在编译时嵌入 SWF 文件，也可在运行时作为外部文件加载。无论采用上述哪一种方式，都要在 ActionScript 中访问着色器，方法是先创建一个 **Shader** 对象，然后将其链接到着色器字节代码。

- 着色器输入：一种复杂的输入，通常为位图图像数据，该数据提供给着色器供其计算之用。对于着色器中定义的每个输入变量，着色器的整个执行过程都使用单一变量值（即，一个图像或一组二进制数据）。
- 着色器参数：提供给着色器供其计算之用的单一值（或限定的值集合）。着色器的每次执行中都会定义各个参数值，该值在着色器的整个执行过程中保持不变。

完成本章中的示例

学习本章的过程中，您可能想要测试所提供的示例代码清单。由于本章介绍的是创建和操作可视内容，所以测试代码涉及到运行代码以及在创建的 SWF 中查看结果。所有示例都使用绘图 API 创建内容，这类 API 使用着色器效果，或由该效果进行修改。

大多数示例代码清单都包含两个部分。一部分是此示例中所用着色器的 Pixel Bender 源代码。您必须首先使用 Pixel Bender 工具包将源代码编译成 Pixel Bender 字节代码文件。请遵循以下步骤创建 Pixel Bender 字节代码文件：

- 1 打开 Adobe Pixel Bender 工具包。如有必要，从“生成”菜单中选择“打开 Flash Player 警告和错误”。
- 2 复制 Pixel Bender 代码清单，然后粘贴到 Pixel Bender 工具包的代码编辑器窗格中。
- 3 从“文件”菜单中，选择“为 Flash Player 导出内核滤镜”。
- 4 将 Pixel Bender 字节代码文件保存到 Flash 文档所在的目录。文件名应与示例说明中指定的名称一致。

每个示例的 ActionScript 部分都编写为一个类文件。测试示例：

- 1 创建一个空的 Flash 文档并将它保存到您的计算机上。
- 2 创建一个新的 ActionScript 文件，并将它保存到 Flash 文档所在的目录中。文件名应与代码清单中的类的名称一致。例如，如果代码清单定义了一个名为 **MyApplication** 的类，则使用 **MyApplication.as** 名称保存 ActionScript 文件。
- 3 将代码清单复制到 ActionScript 文件中并保存该文件。
- 4 在 Flash 文档中，单击舞台或工作区的空白部分，以激活文档的“属性”检查器。
- 5 在“属性”检查器的“文档类”字段中，输入您从文本中复制的 ActionScript 类的名称。
- 6 使用“控制”>“测试影片”来运行程序

您将在预览窗口中看到示例的结果。

这些示例代码清单测试方法在第 29 页的“[测试本章内的示例代码清单](#)”中有更详细的介绍。

加载或嵌入着色器

在 ActionScript 中使用 Pixel Bender 着色器的第一步是在 ActionScript 代码中访问着色器。因为着色器是用 Adobe Pixel Bender 工具包创建并以 Pixel Bender 语言编写的，所以不能在 ActionScript 中直接访问。您需要创建 **Shader** 类的一个实例，用于向 ActionScript 表示 Pixel Bender 着色器。通过 **Shader** 对象可以查明着色器的有关信息，如着色器是否需要参数或输入图像值。将 **Shader** 对象传递给其它对象即可实际使用着色器。例如，若要将着色器用作滤镜，可将 **Shader** 对象分配给 **ShaderFilter** 对象的 **shader** 属性。若要将着色器用作绘制填充，可将 **Shader** 对象作为参数传递给 **Graphics.beginShaderFill()** 方法。

ActionScript 代码可以通过两种方式访问由 Adobe Pixel Bender 工具包创建的着色器 (.pbj 文件)：

- 在运行时加载：可以使用 **URLLoader** 对象将着色器文件作为外部资源进行加载。这种方法类似于加载外部资源（如文本文件）。下面的示例演示如何在运行时加载着色器字节码文件并将其链接到一个 **Shader** 实例：

```
var loader:URLLoader = new URLLoader();
loader.dataFormat = URLLoaderDataFormat.BINARY;
loader.addEventListener(Event.COMPLETE, onLoadComplete);
loader.load(new URLRequest("myShader.pbj"));

var shader:Shader;

function onLoadComplete(event:Event):void {
    // Create a new shader and set the loaded data as its bytecode
    shader = new Shader();
    shader.byteCode = loader.data;

    // You can also pass the bytecode to the Shader() constructor like this:
    // shader = new Shader(loader.data);

    // do something with the shader
}
```

- 嵌入在 SWF 文件中：使用 **[Embed]** 元数据标签可以在编译时将着色器文件嵌入在 SWF 文件中。只有在使用 Flex SDK 编译 SWF 文件时，**[Embed]** 元数据标签才可用。**[Embed]** 标签的 **source** 参数指向着色器文件，其 **mimeType** 参数为 "application/octet-stream"，如此例中所示：

```
[Embed(source="myShader.pbj", mimeType="application/octet-stream")]
var MyShaderClass:Class;

// ...

// create a shader and set the embedded shader as its bytecode
var shaderShader = new Shader();
shaderShader.byteCode = new MyShaderClass();

// You can also pass the bytecode to the Shader() constructor like this:
// var shader:Shader = new Shader(new MyShaderClass());

// do something with the shader
```

在任何一种情况下，都可以将原始着色器字节码 (**URLLoader.data** 属性或 **[Embed]** 数据类的实例) 链接到 **Shader** 实例。如以上示例所示，可以通过两种方式将字节码分配给 **Shader** 实例。可以将着色器字节代码作为参数传递到 **Shader()** 构造函数。或者，可以将其设置为 **Shader** 实例的 **byteCode** 属性。

在创建 Pixel Bender 着色器并将其链接到 **Shader** 对象后，即可使用着色器以多种方式创建效果。您可以将着色器用作滤镜、混合模式、位图填充，也可以用于位图或其它数据的独立处理。您还可以使用 **Shader** 对象的 **data** 属性访问着色器的元数据、指定输入图像以及设置参数值。

访问着色器元数据

在创建 Pixel Bender 着色器内核时，作者可以在 Pixel Bender 源代码中指定着色器的相关元数据。在 ActionScript 中使用着色器时，可以检查着色器和提取其元数据。

在创建 Shader 实例并将其链接到 Pixel Bender 着色器时，会创建一个包含着色器相关数据的 **ShaderData** 对象并将该对象存储在 **Shader** 对象的 **data** 属性中。**ShaderData** 类不定义自身的任何属性。但是，对于在着色器源代码中定义的每个元数据值，都会在运行时将一个属性动态添加到 **ShaderData** 对象。为每个属性提供的名称与在元数据中指定的名称相同。例如，假设 Pixel Bender 着色器的源代码包括下面的元数据定义：

```
namespace : "Adobe::Example";
vendor : "Bob Jones";
version : 1;
description : "Creates a version of the specified image with the specified brightness.";
```

则为该着色器创建 **ShaderData** 对象时将使用以下属性和值：

- **namespace** (字符串) : "Adobe::Example"
- **vendor** (字符串) : "Bob Jones"
- **version** (字符串) : "1"
- **description** (字符串) : "Creates a version of the specified image with the specified brightness"

因为元数据属性是动态添加到 **ShaderData** 对象的，所以可以使用 **for..in** 循环检查 **ShaderData** 对象。通过这种方法，可以确定着色器是否具有元数据以及元数据是何值。除了元数据属性之外，**ShaderData** 对象还可能具有表示着色器中定义的输入和参数的属性。在使用 **for..in** 循环检查 **ShaderData** 对象时，检查每个属性的数据类型以确定属性是输入 (**ShaderInput** 实例)、参数 (**ShaderParameter** 实例) 还是元数据值 (**String** 实例)。下面的示例演示如何使用 **for..in** 循环检查着色器的 **data** 属性的动态属性。每个元数据值都会添加到一个名为 **metadata** 的 **Vector** 实例。请注意，此示例假设已创建了名为 **myShader** 的 **Shader** 实例：

```
var shaderData:ShaderData = myShader.data;
var metadata:Vector.<String> = new Vector.<String>();

for (var prop:String in shaderData)
{
    if (!(shaderData[prop] is ShaderInput) && !(shaderData[prop] is ShaderParameter))
    {
        metadata[metadata.length] = shaderData[prop];
    }
}

// do something with the metadata
```

有关增加着色器输入和参数提取的此示例版本，请参阅第 323 页的“[识别着色器输入和参数](#)”。有关输入和参数属性的详细信息，请参阅第 322 页的“[指定着色器输入和参数值](#)”。

指定着色器输入和参数值

许多 Pixel Bender 着色器定义为使用一个或多个在着色器处理中使用的输入图像。例如，一种常见情况是，着色器接受一个源图像，然后输出应用了特定效果的该图像。根据着色器的使用方式，可以自动指定输入值，也可能需要显式提供值。同样，许多着色器指定用于自定义着色器输出的参数。在使用着色器之前，还必须为每个参数显式设置一个值。

使用 **Shader** 对象的 **data** 属性可以设置着色器输入和参数，还可以确定特定着色器是否需要输入或参数。**data** 属性是一个 **ShaderData** 实例。

识别着色器输入和参数

指定着色器输入和参数值的第一步是确定要使用的特定着色器是否需要输入图像或参数。每个 **Shader** 实例都有一个包含 **ShaderData** 对象的 **data** 属性。如果着色器定义了任何输入或参数，则这些输入或参数可作为该 **ShaderData** 对象的属性进行访问。属性的名称与在着色器源代码中为输入和参数指定的名称相匹配。例如，如果着色器定义了一个名为 **src** 的输入，则 **ShaderData** 对象具有一个名为 **src** 的属性用于表示该输入。每个表示输入的属性都是一个 **ShaderInput** 实例，而每个表示参数的属性都是一个 **ShaderParameter** 实例。

理想情况下，着色器的创作者会提供着色器文档，用于指示着色器所需的输入图像值和参数、其所表示的内容以及正确的值等等。

但是，如果着色器不带有文档（并且您没有其源代码），则可以检查着色器数据以确定输入和参数。表示输入和参数的属性会动态添加到 **ShaderData** 对象。因此，可以使用 **for..in** 循环检查 **ShaderData** 对象，以确定其关联着色器是否定义了输入或参数。如第 322 页的“[访问着色器元数据](#)”中所述，为着色器定义的任何元数据值还可以作为添加到 **Shader.data** 属性的动态属性进行访问。在使用这种方法识别着色器输入和参数时，请检查动态属性的数据类型。如果属性是 **ShaderInput** 实例，则表示输入。如果是 **ShaderParameter** 实例，则表示参数。其它情况下，则是元数据值。下面的示例演示如何使用 **for..in** 循环检查着色器的 **data** 属性的动态属性。每个输入（**ShaderInput** 对象）都会添加到一个名为 **inputs** 的 **Vector** 实例。每个参数（**ShaderParameter** 对象）都会添加到一个名为 **parameters** 的 **Vector** 实例。最后，所有元数据属性都会添加到一个名为 **metadata** 的 **Vector** 实例。请注意，此示例假设已创建了名为 **myShader** 的 **Shader** 实例：

```
var shaderData:ShaderData = myShader.data;
var inputs:Vector.<ShaderInput> = new Vector.<ShaderInput>();
var parameters:Vector.<ShaderParameter> = new Vector.<ShaderParameter>();
var metadata:Vector.<String> = new Vector.<String>();

for (var prop:String in shaderData)
{
    if (shaderData[prop] is ShaderInput)
    {
        inputs[inputs.length] = shaderData[prop];
    }
    else if (shaderData[prop] is ShaderParameter)
    {
        parameters[parameters.length] = shaderData[prop];
    }
    else
    {
        metadata[metadata.length] = shaderData[prop];
    }
}

// do something with the inputs or properties
```

指定着色器输入值

许多着色器需要一个或多个在着色器处理中使用的输入图像。但是，在很多情况下，输入是在使用 **Shader** 对象时自动指定的。例如，假定着色器需要一个输入，并且该着色器用作滤镜。在滤镜应用于显示对象或 **BitmapData** 对象时，该对象会自动设置为输入。在这种情况下，无需显式设置输入值。

但是，在某些情况下，特别是在着色器定义多个输入时，一定要显式设置输入值。在 ActionScript 中，在着色器中定义的每个输入都是由一个 **ShaderInput** 对象表示的。如第 323 页的“[识别着色器输入和参数](#)”中所述，**ShaderInput** 对象是 **Shader** 对象的 **data** 属性中的 **ShaderData** 实例的属性。例如，假定着色器定义一个名为 **src** 的输入，并且该着色器链接到一个名为 **myShader** 的 **Shader** 对象。在这种情况下，可使用下面的标识符访问对应于 **src** 输入的 **ShaderInput** 对象：

```
myShader.data.src
```

每个 `ShaderInput` 对象都有一个 `input` 属性，该属性用于设置输入值。将 `input` 属性设置为 `BitmapData` 实例可指定图像数据。此外，还可以将 `input` 属性设置为 `BitmapData` 或 `Vector.<Number>` 实例以指定二进制或数字数据。有关使用 `BitmapData` 或 `Vector.<Number>` 实例作为输入的详细信息和限制，请参阅语言参考中的 `ShaderInput.input` 列表。

除了 `input` 属性之外，`ShaderInput` 对象还具有可用于确定输入所需图像类型的属性。这些属性包括 `width`、`height` 和 `channels` 属性。每个 `ShaderInput` 对象还有一个 `index` 属性，该属性用于确定是否必须为输入提供显式值。如果着色器需要的输入数量大于自动设置的输入数量，则需要为这些输入设置值。有关使用着色器的不同方式，以及是否自动设置输入值的详细信息，请参阅第 327 页的“[使用着色器](#)”。

指定着色器参数值

某些着色器定义了参数值，着色器在创建其结果时会使用这些参数值。例如，更改图像亮度的着色器可能会指定一个亮度参数，该参数确定操作影响亮度的程度。根据着色器中的参数定义，着色器中定义的单个参数可能需要单个值或多个值。在 ActionScript 中，在着色器中定义的每个参数都由一个 `ShaderParameter` 对象表示。如第 323 页的“[识别着色器输入和参数](#)”中所述，`ShaderParameter` 对象是 `Shader` 对象的 `data` 属性中的 `ShaderData` 实例的属性。例如，假定着色器定义一个名为 `brightness` 的参数，并且该着色器由一个名为 `myShader` 的 `Shader` 对象表示。在这种情况下，可使用下面的标识符访问对应于 `brightness` 参数的 `ShaderParameter`：

```
myShader.data.brightness
```

若要为该参数设置一个（或多个）值，请创建包含这个值（或这些值）的 ActionScript 数组，并将该数组分配给 `ShaderParameter` 对象的 `value` 属性。`value` 属性定义为 `Array` 实例，是因为单个着色器参数可能需要多个值。即使着色器参数只需要一个值，也必须将该值包含在 `Array` 对象中才能将它分配给 `ShaderParameter.value` 属性。下面的代码演示如何将单个值设置为 `value` 属性：

```
myShader.data.brightness.value = [75];
```

如果着色器的 Pixel Bender 源代码为参数定义默认值，则在创建 `Shader` 对象时，会创建包含默认值的数组，并将其分配给 `ShaderParameter` 对象的 `value` 属性。将数组分配给 `value` 属性之后（包括该数组是默认数组的情况），可以通过更改数组元素的值来更改参数值。您不需要创建新数组并将它分配给 `value` 属性。

下面的示例演示如何在 ActionScript 中设置着色器的参数值。在此示例中，着色器定义一个名为 `color` 的参数。在 Pixel Bender 源代码中，`color` 参数声明为 `float4` 变量，这意味着该参数是包含四个浮点数的数组。在该示例中，`color` 参数值会连续更改，该参数每次更改时，都会使用着色器在屏幕上绘制一个有色的矩形。产生的结果是具有动画效果的颜色更改。

注：此示例的代码由 Ryan Taylor 编写。感谢 Ryan 分享此示例。若要查看 Ryan 的个人资料并阅读他的文章，请访问 www.boostworthy.com/。

这段 ActionScript 代码以下面三个方法为中心：

- `init()`：在 `init()` 方法中，代码加载包含着色器的 Pixel Bender 字节代码文件。加载文件后，调用 `onLoadComplete()` 方法。
- `onLoadComplete()`：在 `onLoadComplete()` 方法中，代码创建名为 `shader` 的 `Shader` 对象。同时还创建了名为 `texture` 的 `Sprite` 实例。在 `renderShader()` 方法中，代码将着色器结果逐帧绘制在 `texture` 中。
- `onEnterFrame()`：`onEnterFrame()` 方法按每帧一次的频率进行调用，用于创建动画效果。在此方法中，代码将着色器参数值设置为新颜色，然后调用 `renderShader()` 方法将着色器结果绘制为矩形。
- `renderShader()`：在 `renderShader()` 方法中，代码调用 `Graphics.beginShaderFill()` 方法来指定着色器填充。然后，绘制一个矩形，该矩形的填充由着色器输出（生成的颜色）定义。有关以此方法使用着色器的详细信息，请参阅第 327 页的“[使用着色器作为绘制填充](#)”。

下面是此示例的 ActionScript 代码。使用此类作为 Flex 中仅包含 ActionScript 项目的主应用程序类，或者作为 Flash 创作工具中 FLA 文件的文档类：

```
package
{
    import flash.display.Shader;
    import flash.display.Sprite;
    import flash.events.Event;
    import flash.net.URLLoader;
    import flash.net.URLLoaderDataFormat;
    import flash.net.URLRequest;

    public class ColorFilterExample extends Sprite
    {
        private const DELTA_OFFSET:Number = Math.PI * 0.5;
        private var loader:URLLoader;
        private var shader:Shader;
        private var texture:Sprite;
        private var delta:Number = 0;

        public function ColorFilterExample()
        {
            init();
        }

        private function init():void
        {
            loader = new URLLoader();
            loader.dataFormat = URLLoaderDataFormat.BINARY;
            loader.addEventListener(Event.COMPLETE, onLoadComplete);
            loader.load(new URLRequest("ColorFilter.pbj"));
        }

        private function onLoadComplete(event:Event):void
        {
            shader = new Shader(loader.data);

            shader.data.point1.value = [topMiddle.x, topMiddle.y];
            shader.data.point2.value = [bottomLeft.x, bottomLeft.y];
            shader.data.point3.value = [bottomRight.x, bottomRight.y];
            texture = new Sprite();

            addChild(texture);

            addEventListener(Event.ENTER_FRAME, onEnterFrame);
        }
        private function onEnterFrame(event:Event):void
        {
```

```

        shader.data.color.value[0] = 0.5 + Math.cos(delta - DELTA_OFFSET) * 0.5;
        shader.data.color.value[1] = 0.5 + Math.cos(delta) * 0.5;
        shader.data.color.value[2] = 0.5 + Math.cos(delta + DELTA_OFFSET) * 0.5;
        // The alpha channel value (index 3) is set to 1 by the kernel's default
        // value. This value doesn't need to change.

        delta += 0.1;

        renderShader();
    }

    private function renderShader():void
    {
        texture:graphics.clear();
        texture.graphics.beginShaderFill(shader);
        texture.graphics.drawRect(0, 0, stage.stageWidth, stage.stageHeight);
        texture.graphics.endFill();
    }
}
}

```

下面是 ColorFilter 着色器内核的源代码，用于创建“ColorFilter.pbj”Pixel Bender 字节代码文件：

```

<languageVersion : 1.0;>
kernel ColorFilter
<
    namespace : "boostworthy::Example";
    vendor : "Ryan Taylor";
    version : 1;
    description : "Creates an image where every pixel has the specified color value.";
>
{
    output pixel4 result;

    parameter float4 color
    <
        minValue:float4(0, 0, 0, 0);
        maxValue:float4(1, 1, 1, 1);
        defaultValue:float4(0, 0, 0, 1);
    >

    void evaluatePixel()
    {
        result = color;
    }
}

```

如果所使用的着色器的参数没有文档说明，则通过检查 **ShaderParameter** 对象的 **type** 属性，便可确定数组中必须包含的元素数量及其类型。**type** 属性指示着色器本身定义的参数数据类型。有关每个参数类型需要的元素数量和类型的列表，请参阅语言参考中的 **ShaderParameter.value** 属性列表。

每个 **ShaderParameter** 对象还有一个 **index** 属性，该属性指示参数在着色器参数顺序中的位置。除了这些属性之外，**ShaderParameter** 对象还可以具有包含由着色器创作者提供的元数据值的其它属性。例如，创作者可以指定元数据值，如参数的最小值、最大值和默认值。创作者所指定的所有元数据值都会作为动态属性添加到 **ShaderParameter** 对象中。若要检查这些属性，请使用 **for.in** 循环来循环访问 **ShaderParameter** 对象的动态属性，以识别其元数据。下面的示例演示如何使用 **for.in** 循环来识别 **ShaderParameter** 对象的元数据。每个元数据值都会添加到一个名为 **metadata** 的 **Vector** 实例。请注意，此示例假定已创建了一个名为 **myShader** 的 **Shader** 实例，并且该实例具有一个名为 **brightness** 的参数：

```

var brightness:ShaderParameter = myShader.data.brightness;
var metadata:Vector.<String> = new Vector.<String>();

for (var prop:String in brightness)
{
    if (brightness[prop] is String)
    {
        metadata[metadata.length] = brightness[prop];
    }
}

// do something with the metadata

```

使用着色器

只要 Pixel Bender 着色器可以在 ActionScript 中用作 **Shader** 对象，就可以通过多种方式使用：

- 着色器绘制填充：着色器定义使用绘图 API 绘制的形状的填充部分。
- 混合模式：着色器定义两个重叠的显示对象之间的混合
- 滤镜：着色器定义用于修改可视内容外观的滤镜。
- 独立着色器处理：着色器处理在不指定输出的目标用途的情况下运行。着色器也可以在后台运行，结果在处理完成时可用。这种方法可用于生成位图数据，还可用于处理非可视数据。

使用着色器作为绘制填充

在使用着色器创建绘制填充时，您将使用绘图 API 方法来创建矢量形状。正如使用绘图 API 可将任何位图图像用作位图填充一样，着色器的输出将用于填充该形状。若要创建着色器填充，可在代码中开始绘制形状的位置处，调用 **Graphics** 对象的 **beginShaderFill()** 方法。将 **Shader** 对象作为第一个参数传递给 **beginShaderFill()** 方法，如以下清单所示：


```

var canvas:Sprite = new Sprite();
canvas.graphics.beginShaderFill(myShader);
canvas.graphics.drawRect(10, 10, 150, 150);
canvas.graphics.endFill();
// add canvas to the display list to see the result

```

在使用着色器作为绘制填充时，您将设置着色器需要的所有输入图像值和参数值。

下面的示例演示如何使用着色器作为绘制填充。在此示例中，着色器创建了一个三点渐变。此渐变具有三种颜色，分别位于三角形的三个顶点，之间的颜色为渐变混和。此外，颜色进行转动，产生旋转色的动画效果。



注：此示例的代码由 Petri Leskinen 编写。感谢 Petri 分享此示例。若要查看 Petri 的更多示例和教程，请访问 <http://pixelero.wordpress.com/>。

这段 ActionScript 代码用到三个方法：

- `init()`：应用程序加载时调用 `init()` 方法。在此方法中，代码为代表三角形三个顶点的 `Point` 对象设置初始值。此外还创建名为 `canvas` 的 `Sprite` 实例。稍后，在 `updateShaderFill()` 方法中，代码将着色器结果逐帧绘制在 `canvas` 中。最后，代码加载着色器字节代码文件。
- `onLoadComplete()`：在 `onLoadComplete()` 方法中，代码创建名为 `shader` 的 `Shader` 对象。此外还设置初始参数值。最后，代码添加 `updateShaderFill()` 方法作为 `enterFrame` 事件的侦听器，表示逐帧调用该方法来创建动画效果。
- `updateShaderFill()`：`updateShaderFill()` 方法是逐帧调用的，用于创建动画效果。在此方法中，代码计算并设置着色器的参数值。然后，代码调用 `beginShaderFill()` 方法来创建着色器填充，调用其它绘图 API 方法在三角形中绘制着色器结果。

下面是此示例的 ActionScript 代码。使用此类作为 Flex 中仅包含 ActionScript 项目的主应用程序类，或者作为 Flash 创作工具中 FLA 文件的文档类：

```
package
{
    import flash.display.Shader;
    import flash.display.Sprite;
    import flash.events.Event;
    import flash.geom.Point;
    import flash.net.URLLoader;
    import flash.net.URLLoaderDataFormat;
    import flash.net.URLRequest;

    public class ThreePointGradient extends Sprite
    {
        private var canvas:Sprite;
        private var shader:Shader;
        private var loader:URLLoader;

        private var topMiddle:Point;
        private var bottomLeft:Point;
        private var bottomRight:Point;

        private var colorAngle:Number = 0.0;
        private const d120:Number = 120 / 180 * Math.PI; // 120 degrees in radians

        public function ThreePointGradient()
        {
            init();
        }

        private function init():void
        {
            canvas = new Sprite();
            addChild(canvas);

            var size:int = 400;
            topMiddle = new Point(size / 2, 10);
            bottomLeft = new Point(0, size - 10);
            bottomRight = new Point(size, size - 10);

            loader = new URLLoader();
            loader.dataFormat = URLLoaderDataFormat.BINARY;
            loader.addEventListener(Event.COMPLETE, onLoadComplete);
            loader.load(new URLRequest("ThreePointGradient.pbj"));
        }
    }
}
```

```

private function onLoadComplete(event:Event):void
{
    shader = new Shader(loader.data);

    shader.data.point1.value = [topMiddle.x, topMiddle.y];
    shader.data.point2.value = [bottomLeft.x, bottomLeft.y];
    shader.data.point3.value = [bottomRight.x, bottomRight.y];

    addEventListener(Event.ENTER_FRAME, updateShaderFill);
}

private function updateShaderFill(event:Event):void
{
    colorAngle += .06;

    var c1:Number = 1 / 3 + 2 / 3 * Math.cos(colorAngle);
    var c2:Number = 1 / 3 + 2 / 3 * Math.cos(colorAngle + d120);
    var c3:Number = 1 / 3 + 2 / 3 * Math.cos(colorAngle - d120);

    shader.data.color1.value = [c1, c2, c3, 1.0];
    shader.data.color2.value = [c3, c1, c2, 1.0];
    shader.data.color3.value = [c2, c3, c1, 1.0];

    canvas.graphics.clear();
    canvas.graphics.beginShaderFill(shader);

    canvas.graphics.moveTo(topMiddle.x, topMiddle.y);
    canvas.graphics.lineTo(bottomLeft.x, bottomLeft.y);
    canvas.graphics.lineTo(bottomRight.x, bottomLeft.y);

    canvas.graphics.endFill();
}
}
}

```

下面是 ThreePointGradient 着色器内核的源代码，用于创建“ThreePointGradient.pbj”Pixel Bender 字节代码文件：

```

<languageVersion : 1.0;>
kernel ThreePointGradient
<
namespace : "Petri Leskinen::Example";
vendor : "Petri Leskinen";
version : 1;
description : "Creates a gradient fill using three specified points and colors.";
>
{
    parameter float2 point1 // coordinates of the first point
    <
        minValue:float2(0, 0);
        maxValue:float2(4000, 4000);
        defaultValue:float2(0, 0);
    >

    parameter float4 color1 // color at the first point, opaque red by default
    <
        defaultValue:float4(1.0, 0.0, 0.0, 1.0);
    >

    parameter float2 point2 // coordinates of the second point
    <
        minValue:float2(0, 0);
        maxValue:float2(4000, 4000);
    >
}

```

```

        defaultValue:float2(0, 500);
    >

parameter float4 color2 // color at the second point, opaque green by default
<
    defaultValue:float4(0.0, 1.0, 0.0, 1.0);
>

parameter float2 point3 // coordinates of the third point
<
    minValue:float2(0, 0);
    maxValue:float2(4000, 4000);
    defaultValue:float2(0, 500);
>

parameter float4 color3 // color at the third point, opaque blue by default
<
    defaultValue:float4(0.0, 0.0, 1.0, 1.0);
>

output pixel4 dst;

void evaluatePixel()
{
    float2 d2 = point2 - point1;
    float2 d3 = point3 - point1;

    // transformation to a new coordinate system
    // transforms point 1 to origin, point2 to (1, 0), and point3 to (0, 1)
    float2x2 mtrx = float2x2(d3.y, -d2.y, -d3.x, d2.x) / (d2.x * d3.y - d3.x * d2.y);
    float2 pNew = mtrx * (outCoord() - point1);

    // repeat the edge colors on the outside
    pNew.xy = clamp(pNew.xy, 0.0, 1.0); // set the range to 0.0 ... 1.0

    // interpolating the output color or alpha value
    dst = mix(mix(color1, color2, pNew.x), color3, pNew.y);
}
}

```

有关使用绘图 API 绘制形状的详细信息，请参阅第 265 页的“[使用绘图 API](#)”。

使用着色器作为混合模式

使用着色器作为混和模式与使用其它混和模式类似。着色器定义的外观取决于将两个显示对象混和在一起的视觉效果。若要将着色器用于混和模式，请将 **Shader** 对象指派给前景显示对象的 **blendShader** 属性。如果为 **blendShader** 属性指定非 null 值，显示对象的 **blendMode** 属性将自动设置为 **BlendMode.SHADER**。下面的清单演示如何使用着色器作为混和模式。注意，此示例假定存在名为 **foreground** 的显示对象。该对象与其它显示内容包含在显示列表的同一父级中，而 **foreground** 与其它内容重叠：

```
foreground.blendShader = myShader;
```

使用着色器作为混和模式时，着色器必须由至少两个输入定义。如示例所示，您未在代码中设置输入值。而是将两个混和后的图像自动用作着色器的输入。前景图像设置为第二个图像。（此显示对象便是要对其应用混和模式的对象。）背景图像由前景图像边框后的所有像素组合而成。背景图像设置为第一个输入图像。如果所用着色器要求两个以上的输入，则还需为前两个之外的其它输入提供值。


下面的示例演示如何使用着色器作为混和模式。此示例使用基于亮度的加亮混和模式。混和的结果是以任一混和对象中最亮的像素值显示该像素。

注：此示例的代码由 Mario Klingemann 编写。感谢 Mario 分享此示例。若要查看 Mario 的更多作品及阅读他的文章，请访问 www.quasimondo.com/。


这段重要的 ActionScript 代码用到下面两个方法：

- `init()`: 应用程序加载时调用 `init()` 方法。在此方法中，代码加载着色器字节代码文件。
- `onLoadComplete()`: 在 `onLoadComplete()` 方法中，代码创建名为 `shader` 的 `Shader` 对象。然后绘制三个对象。第一个对象 `backdrop` 是混和对象后的深灰色背景。第二个对象 `backgroundShape` 是绿色渐变椭圆。第三个对象 `foregroundShape` 是橙色渐变椭圆。

`foregroundShape` 椭圆是混和的前景对象。混和的背景图像由部分 `backdrop` 和部分 `backgroundShape` 构成，并叠加了 `foregroundShape` 对象的边框。`foregroundShape` 对象是位于显示列表最前面的对象。它与 `backgroundShape` 部分重叠，与 `backdrop` 全部重叠。由于上述重叠的存在，如果不应用混和模式，橙色椭圆 (`foregroundShape`) 将全部显示，并遮盖住绿色椭圆 (`backgroundShape`) 的一部分：



而在应用混和模式之后，绿色椭圆的较亮部分会“穿透”前景显示出来，因为该部分比遮盖它的那部分 `foregroundShape` 更亮。



下面是此示例的 ActionScript 代码。使用此类作为 Flex 中仅包含 ActionScript 项目的主应用程序类，或者作为 Flash 创作工具中 FLA 文件的文档类：

```
package
{
    import flash.display.BlendMode;
    import flash.display.GradientType;
    import flash.display.Graphics;
    import flash.display.Shader;
    import flash.display.Shape;
    import flash.display.Sprite;
    import flash.events.Event;
    import flash.geom.Matrix;
    import flash.net.URLLoader;
    import flash.net.URLLoaderDataFormat;
    import flash.net.URLRequest;

    public class LumaLighten extends Sprite
    {
        private var shader:Shader;
        private var loader:URLLoader;

        public function LumaLighten()
        {
            init();
        }

        private function init():void
        {
            loader = new URLLoader();
            loader.dataFormat = URLLoaderDataFormat.BINARY;
            loader.addEventListener(Event.COMPLETE, onLoadComplete);
            loader.load(new URLRequest("LumaLighten.pbj"));
        }

        private function onLoadComplete(event:Event):void
        {
            shader = new Shader(loader.data);

            var backdrop:Shape = new Shape();
            var g0:Graphics = backdrop.graphics;
            g0.beginFill(0x303030);
            g0.drawRect(0, 0, 400, 200);
            g0.endFill();
            addChild(backdrop);

            var backgroundShape:Shape = new Shape();
            var g1:Graphics = backgroundShape.graphics;
            var c1:Array = [0x336600, 0x80ff00];
            var a1:Array = [255, 255];
            var r1:Array = [100, 255];
            var m1:Matrix = new Matrix();
            m1.createGradientBox(300, 200);
            g1.beginGradientFill(GradientType.LINEAR, c1, a1, r1, m1);
            g1.drawEllipse(0, 0, 300, 200);
        }
    }
}
```

```

g1.endFill();
addChild(backgroundShape);

var foregroundShape:Shape = new Shape();
var g2:Graphics = foregroundShape.graphics;
var c2:Array = [0xff8000, 0x663300];
var a2:Array = [255, 255];
var r2:Array = [100, 255];
var m2:Matrix = new Matrix();
m2.createGradientBox(300, 200);
g2.beginGradientFill(GradientType.LINEAR, c2, a2, r2, m2);
g2.drawEllipse(100, 0, 300, 200);
g2.endFill();
addChild(foregroundShape);

foregroundShape.blendShader = shader;
foregroundShape.blendMode = BlendMode.SHADER;
}
}
}

```

下面是 LumaLighten 着色器内核的源代码，用于创建“LumaLighten.pbj”Pixel Bender 字节代码文件：

```

<languageVersion : 1.0;>
kernel LumaLighten
<
namespace : "com.quasimondo.blendModes";
vendor : "Quasimondo.com";
version : 1;
description : "Luminance based lighten blend mode";
>
{
    input image4 background;
    input image4 foreground;

    output pixel4 dst;

    const float3 LUMA = float3(0.212671, 0.715160, 0.072169);

    void evaluatePixel()
    {
        float4 a = sampleNearest(foreground, outCoord());
        float4 b = sampleNearest(background, outCoord());
        float luma_a = a.r * LUMA.r + a.g * LUMA.g + a.b * LUMA.b;
        float luma_b = b.r * LUMA.r + b.g * LUMA.g + b.b * LUMA.b;

        dst = luma_a > luma_b ? a : b;
    }
}

```

有关使用混和模式的详细信息，请参阅第 250 页的“[应用混合模式](#)”。

使用着色器作为滤镜

使用着色器作为滤镜与在 ActionScript 中使用任何其它滤镜类似。使用着色器作为滤镜时，过滤出的图像（显示对象或 BitmapData 对象）将传递给着色器。着色器使用输入图像来创建滤镜输出，该输出通常为原始图像经过修改的版本。如果过滤出的对象是显示对象，着色器输出将显示在屏幕上，代替过滤出的显示对象。如果过滤出的对象是 BitmapData 对象，着色器输出将成为 BitmapData 对象的内容，并调用该对象的 applyFilter() 方法。

若要使用着色器作为滤镜，您首先要创建 `Shader` 对象，如第 321 页的“[加载或嵌入着色器](#)”中所述。接下来，您需要创建链接到该 `Shader` 对象的 `ShaderFilter` 对象。`ShaderFilter` 对象便是将应用到所过滤对象的滤镜。将滤镜应用于对象的方式与应用任何滤镜的方式相同。将其传递给显示对象的 `filters` 属性，或者对 `BitmapData` 对象调用 `applyFilter()` 方法。例如，下面的代码创建 `ShaderFilter` 对象，并将此滤镜应用到名为 `homeButton` 的显示对象。

```
var myFilter:ShaderFilter = new ShaderFilter(myShader);
homeButton.filters = [myFilter];
```

在使用着色器作为滤镜时，着色器必须由至少一个输入定义。如示例所示，您未在代码中设置输入值。而是将过滤出的显示对象或 `BitmapData` 对象设置为输入图像。如果所用着色器要求一个以上的输入，则还需为第一个之外的其它输入提供值。


在某些情况下，滤镜会改变原始图像的尺寸。例如，典型的投影效果会添加额外的像素，这些像素组成为图像添加的阴影。在使用改变图像尺寸的着色器时，需设置 `leftExtension`、`rightExtension`、`topExtension` 和 `bottomExtension` 属性，指明所需图像尺寸变化量。

下面的示例演示如何使用着色器作为滤镜。此示例中的滤镜将反转图像红绿蓝三色通道的值。其结果为该图像的“负片”版本。


注：此示例使用的着色器为 Pixel Bender 工具包中附带的 `invertRGB.pbk` Pixel Bender 内核。您可以从 Pixel Bender 工具包安装目录中加载此内核的源代码。编译源代码，然后将字节代码文件保存到源代码所在的目录。

这段重要的 ActionScript 代码用到下面两个方法：

- `init()`：应用程序加载时调用 `init()` 方法。在此方法中，代码加载着色器字节代码文件。
- `onLoadComplete()`：在 `onLoadComplete()` 方法中，代码创建名为 `shader` 的 `Shader` 对象。然后创建并绘制名为 `target` 的对象的内容。`target` 对象是填充有线性渐变色的矩形：左边是红色，中间是黄绿色，右边是淡蓝色。未过滤的对象看上去是这样：



应用了滤镜后，颜色反转，矩形变成这样：



此示例使用的着色器为 Pixel Bender 工具包中附带的“`invertRGB.pbk`”Pixel Bender 范例内核。源代码位于 Pixel Bender 工具包安装目录的“`invertRGB.pbk`”文件中。编译源代码，然后以文件名“`invertRGB.pbj`”将字节代码文件保存在 ActionScript 源代码所在的目录中。

下面是此示例的 ActionScript 代码。使用此类作为 Flex 中仅包含 ActionScript 项目的主应用程序类，或者作为 Flash 创作工具中 FLA 文件的文档类：

```
package
{
    import flash.display.GradientType;
    import flash.display.Graphics;
    import flash.display.Shader;
    import flash.display.Shape;
    import flash.display.Sprite;
    import flash.filters.ShaderFilter;
    import flash.events.Event;
    import flash.geom.Matrix;
    import flash.net.URLLoader;
    import flash.net.URLLoaderDataFormat;
    import flash.net.URLRequest;

    public class InvertRGB extends Sprite
    {
        private var shader:Shader;
        private var loader:URLLoader;

        public function InvertRGB()
        {
            init();
        }

        private function init():void
        {
            loader = new URLLoader();
            loader.dataFormat = URLLoaderDataFormat.BINARY;
            loader.addEventListener(Event.COMPLETE, onLoadComplete);
            loader.load(new URLRequest("invertRGB.pbj"));
        }

        private function onLoadComplete(event:Event):void
        {
            shader = new Shader(loader.data);

            var target:Shape = new Shape();
            addChild(target);

            var g:Graphics = target.graphics;
            var c:Array = [0x990000, 0x445500, 0x007799];
            var a:Array = [255, 255, 255];
            var r:Array = [0, 127, 255];
            var m:Matrix = new Matrix();
            m.createGradientBox(w, h);
            g.beginGradientFill(GradientType.LINEAR, c, a, r, m);
            g.drawRect(10, 10, w, h);
            g.endFill();

            var invertFilter:ShaderFilter = new ShaderFilter(shader);
            target.filters = [invertFilter];
        }
    }
}
```

有关应用滤镜的详细信息，请参阅第 294 页的“[创建和应用滤镜](#)”。

在独立模式下使用着色器

在独立模式下使用着色器时，着色器处理的运行独立于其输出结果的用途。指定要执行的着色器、设置输入值和参数值，及指定用于放置结果数据的对象。在以下两种情形中，可以考虑以独立模式使用着色器：

- 处理非图像数据：在独立模式下，您可以选择将任意二进制数据或数值数据（而非位图图像数据）传递给着色器。除位图图像数据外，您还可以选择将着色器结果以二进制数据或数值数据的形式返回。
- 背景处理：以独立模式运行着色器时，着色器默认为异步运行。这表示，在您的应用程序继续运行的同时，着色器以后台方式运行，并在其处理结束时通知您的代码。您可以使用运行耗时的着色器，它在运行时不会导致应用程序用户界面或其它处理响应迟缓。

使用 `ShaderJob` 对象以独立模式执行着色器。首先创建 `ShaderJob` 对象，并将其链接到代表要执行的着色器的 `Shader` 对象：

```
var job:ShaderJob = new ShaderJob(myShader);
```

接下来，设置着色器需要的所有输入值或参数值。如果着色器在后台运行，还需要为 `ShaderJob` 对象的 `complete` 事件注册一个侦听器。着色器完成其处理时，将调用该侦听器：

```
function completeHandler(event:ShaderEvent):void
{
    // do something with the shader result
}

job.addEventListener(ShaderEvent.COMPLETE, completeHandler);
```

接下来，创建着色器操作完成时，向其中写入操作结果的对象。将该对象指派给 `ShaderJob` 对象的 `target` 属性：

```
var jobResult:BitmapData = new BitmapData(100, 75);
job.target = jobResult;
```

如果使用 `ShaderJob` 执行图像处理，则为 `target` 属性指派一个 `BitmapData` 实例。如果要处理二进制数据或数值数据，则指派 `ByteArray` 对象或 `Vector.<Number>` 实例给 `target` 属性。在该情形下，您必须设置 `ShaderJob` 对象的 `width` 和 `height` 属性，以指定输出到 `target` 对象的数据量。

注：您可以一步完成 `ShaderJob` 对象 `shader`、`target`、`width` 和 `height` 属性的设置，方法是将相应的参数传递给 `ShaderJob()` 构造函数，如：`var job:ShaderJob = new ShaderJob(myShader, myTarget, myWidth, myHeight);`

准备好执行着色器时，调用 `ShaderJob` 对象的 `start()` 方法：

```
job.start();
```

默认情况下，调用 `start()` 导致 `ShaderJob` 以异步方式执行。在这种情况下，程序立即继续执行下一行代码，而不等待着色器完成处理。着色器操作完成时，`ShaderJob` 对象调用其 `complete` 事件的侦听器，通知它们操作已完成。在此（即 `complete` 事件侦听器代码中），`target` 对象获得着色器操作结果。

注：也可以不使用 `target` 属性对象，直接从传递给侦听器方法的事件对象处取得着色器结果。该事件对象是一个 `ShaderEvent` 实例。根据设置为 `target` 属性的对象的数据类型，`ShaderEvent` 对象有三个可用于访问结果的属性：`ShaderEvent.bitmapData`、`ShaderEvent.byteArray` 和 `ShaderEvent.vector`。

或者，可以将 `true` 参数传递给 `start()` 方法。在该情形下，着色器操作将同步执行。所有代码（包括与用户界面及所有其它事件的交互）在着色器执行时暂停。着色器完成处理后，`target` 对象包含着色器结果，程序继续执行下一行代码。

```
job.start(true);
```

第 18 章：处理影片剪辑

MovieClip 类是在 Adobe® Flash® CS4 Professional 中创建的动画和影片剪辑元件的核心类。该类除具有显示对象的所有行为和功能外，还具有用于控制影片剪辑的时间轴的其它属性和方法。本章介绍如何使用 ActionScript 来控制影片剪辑播放和动态创建影片剪辑。

影片剪辑基础知识

影片剪辑处理简介

影片剪辑对于使用 Flash 创作工具创建动画内容并想要通过 ActionScript 来控制该内容的人来说是一个重要元素。只要在 Flash 中创建影片剪辑元件，Flash 就会将该元件添加到该 Flash 文档的库中。默认情况下，此元件会成为 **MovieClip** 类，因此具有 MovieClip 类的属性和方法。

在将某个影片剪辑元件的实例放置在舞台上时，如果该影片剪辑具有多个帧，它会自动按其时间轴进行回放，除非使用 ActionScript 更改其回放。此时间轴使 MovieClip 类与其它类区别开来，允许您在 Flash 创作工具中通过补间动画或补间形状来创建动画。相反，对于作为 **Sprite** 类的实例的显示对象，您只需以编程方式更改该对象的值即可创建动画。

在 ActionScript 的早期版本中，MovieClip 类是舞台上所有实例的基类。在 ActionScript 3.0 中，影片剪辑只是可以在屏幕上显示的众多显示对象中的一个。如果使用显示对象时不需要时间轴，则使用 Shape 类或 Sprite 类替代 MovieClip 类可能会提高呈现性能。有关为任务选择合适的显示对象的详细信息，请参阅第 240 页的“[选择 DisplayObject 子类](#)”。

常见的影片剪辑任务

本章介绍以下常见的影片剪辑任务：

- 播放和停止影片剪辑
- 反向播放影片剪辑
- 将播放头移动到影片剪辑时间轴中的特定点
- 在 ActionScript 中处理帧标签
- 在 ActionScript 中访问场景信息
- 使用 ActionScript 创建库影片剪辑元件的实例
- 加载和控制外部 SWF 文件，包括为以前版本的 Flash Player 创建的文件
- 构建一个 ActionScript 系统，用于创建将在运行时加载和使用的图形资源

重要概念和术语

以下参考列表包含将会在本章中使用的重要术语：

- AVM1 SWF：使用 ActionScript 1.0 或 ActionScript 2.0 创建的 SWF 文件，通常以 Flash Player 8 或更早版本为目标播放器。
- AVM2 SWF：使用 ActionScript 3.0 for Adobe Flash Player 9（或更高版本）或 Adobe AIR 创建的 SWF 文件。
- 外部 SWF：单独从项目 SWF 文件创建的 SWF 文件，将加载到项目 SWF 文件中并在该 SWF 文件中回放。

- 帧：时间轴上划分时间的最小单位。与运动图像电影胶片一样，每个帧都类似于动画在特定时间的快照，当快速按顺序播放各个帧时，会产生动画的效果。
- 时间轴：构成影片剪辑动画序列的一系列帧的比喻形式。MovieClip 对象的时间轴等同于 Flash 创作工具中的时间轴。
- 播放头：一个标记，用于标识在给定时刻在时间轴中所处的位置（帧）。

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。由于本章是关于在 ActionScript 中处理影片剪辑的，因此，本章中编写的所有代码清单的出发点几乎都是说明如何操作已经创建并放置在舞台上的影片剪辑元件。测试范例将涉及在 Flash Player 或 AIR 中查看结果，以了解代码对元件的影响。要测试本章中的代码清单，请执行以下操作：

- 1 创建一个空的 Flash 文档。
- 2 在时间轴上选择一个关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 在舞台上创建一个影片剪辑元件实例。例如，绘制一个形状，选中它，选择“修改”>“转换为元件”，并给元件指定一个名称。
- 5 使影片剪辑保持选中，在“属性”检查器中为它指定一个实例名称。名称应与示例代码清单中影片剪辑使用的名称相匹配，例如，如果代码清单操作名为 myMovieClip 的影片剪辑，则您也应将影片剪辑实例命名为 myMovieClip。
- 6 通过“控制”>“测试影片”运行程序。

在屏幕上，您将看到按照代码清单所指定的要求操作影片剪辑的结果。

用于测试示例代码清单的其它方法在第 29 页的“[测试本章内的示例代码清单](#)”中有更详细的介绍。

处理 MovieClip 对象

在发布 SWF 文件时，Flash 会将舞台上的所有影片剪辑元件实例转换为 MovieClip 对象。通过在属性检查器的“实例名称”字段中指定影片剪辑元件的实例名称，您可以在 ActionScript 中使用该元件。在创建 SWF 文件时，Flash 会生成在舞台上创建该 MovieClip 实例的代码并使用该实例名称声明一个变量。如果您已经命名了嵌套在其它已命名影片剪辑内的影片剪辑，则会将这些子级影片剪辑视为父级影片剪辑的属性，这样您便可以使用点语法访问孩子影片剪辑。例如，如果实例名称为 childClip 的影片剪辑嵌套在实例名称为 parentClip 的另一个剪辑内，则可以通过调用以下代码来播放子剪辑的时间轴动画：

```
parentClip.childClip.play();
```

注：在 Flash 创作工具中放到舞台上的子实例无法由父实例构造函数中的代码进行访问，因为在执行代码时尚未在该位置创建这些实例。在访问子实例之前，父实例必须通过代码创建子实例，或者延迟访问用于侦听子实例以调度其 Event.ADDED_TO_STAGE 事件的回调函数。

尽管 ActionScript 2.0 MovieClip 类的一些旧方法和属性仍保持不变，但其它方法和属性已发生了变化。所有前缀为下划线的属性均已被重新命名。例如，_width 和 _height 属性现在分别作为 width 和 height 被访问，而 _xscale 和 _yscale 则作为 scaleX 和 scaleY 被访问。若要查看 MovieClip 类的属性和方法的完整列表，请参阅 ActionScript 3.0 Language and Components Reference.

控制影片剪辑回放

Flash 利用时间轴来形象地表示动画或状态改变。任何使用时间轴的可视元素都必须为 MovieClip 对象或从 MovieClip 类扩展而来。尽管 ActionScript 可控制任何影片剪辑的停止、播放或转至时间轴上的另一点，但不能用于动态创建时间轴或在特定帧添加内容，这项工作仅能使用 Flash 创作工具来完成。

MovieClip 在播放时将以 SWF 文件的帧速率决定的速度沿着其时间轴推进。或者，您也可以通过在 ActionScript 中设置 Stage.frameRate 属性来覆盖此设置。

播放影片剪辑和停止回放

play() 和 stop() 方法允许对时间轴上的影片剪辑进行基本控制。例如，假设舞台上有一个影片剪辑元件，其中包含一个自行车横穿屏幕的动画，其实例名称设置为 bicycle。如果将以下代码附加到主时间轴上的关键帧，

```
bicycle.stop();
```

自行车将不会移动（将不播放其动画）。自行车的移动可以通过其它某种用户交互来开始。例如，如果您有一个名为 startButton 的按扭，则主时间轴上某一关键帧上的以下代码会使单击该按扭时播放该动画：

```
// This function will be called when the button is clicked. It causes the
// bicycle animation to play.
function playAnimation(event:MouseEvent):void
{
    bicycle.play();
}
// Register the function as a listener with the button.
startButton.addEventListener(MouseEvent.CLICK, playAnimation);
```

快进和后退

在影片剪辑中，play() 和 stop() 方法并非是控制播放的唯一方法。也可以使用 nextFrame() 和 prevFrame() 方法手动向前或向后沿时间轴移动播放头。调用这两种方法中的任一方法均会停止回放并分别使播放头向前或向后移动一帧。

使用 play() 方法类似于每次触发影片剪辑对象的 enterFrame 事件时调用 nextFrame()。使用此方法，您可以为 enterFrame 事件创建一个事件侦听器并在侦听器函数中让 bicycle 回到前一帧，从而使 bicycle 影片剪辑向后播放，如下所示：

```
// This function is called when the enterFrame event is triggered, meaning
// it's called once per frame.
function everyFrame(event:Event):void
{
    if (bicycle.currentFrame == 1)
    {
        bicycle.gotoAndStop(bicycle.totalFrames);
    }
    else
    {
        bicycle.prevFrame();
    }
}
bicycle.addEventListener(Event.ENTER_FRAME, everyFrame);
```

在正常回放过程中，如果影片剪辑包含多个帧，播放时将会无限循环播放，也就是说在经过最后一帧后将返回到第 1 帧。使用 prevFrame() 或 nextFrame() 时，不会自动发生此行为（在播放头位于第 1 帧时调用 prevFrame() 不会将播放头移动到最后一个帧）。以上示例中的 if 条件将检查播放头是否已返回至第一帧，并将播放头设置为处于最后一帧前面，从而有效地使影片剪辑向后持续循环播放。

跳到不同帧和使用帧标签

向新帧发送影片剪辑非常简单。调用 gotoAndPlay() 或 gotoAndStop() 将使影片剪辑跳到指定为参数的帧编号。或者，您可以传递一个与帧标签名称匹配的字符串。可以为时间轴上的任何帧分配一个标签。为此，选择时间轴上的某一帧，然后在属性检查器的“帧标签”字段中输入一个名称。

当创建复杂的影片剪辑时，使用帧标签比使用帧编号具有明显优势。当动画中的帧、图层和补间变得很多时，应考虑给重要的帧加上具有解释性说明的标签来表示影片剪辑中的行为转换（例如，“离开”、“行走”或“跑”）。这可提高代码的可读性，同时使代码更加灵活，因为转到添加了标签的帧的 ActionScript 调用是指向单个引用（即“标签”，而不是特定帧编号）的指针。如果您以后决定将动画的特定片段移动到不同的帧，则无需更改 ActionScript 代码，只要将这些帧的相同标签保持在新位置即可。

为便于在代码中表示帧标签，ActionScript 3.0 包括了 **FrameLabel** 类。此类的每个实例均代表一个帧标签，并具有一个 **name** 属性（表示在属性检查器中指定的帧标签的名称）和一个 **frame** 属性（表示该标签在时间轴上所处帧的帧编号）。

为了访问与影片剪辑实例相关联的 **FrameLabel** 实例，**MovieClip** 类包括了两个可直接返回 **FrameLabel** 对象的属性。**currentLabels** 属性返回一个包含影片剪辑整个时间轴上所有 **FrameLabel** 对象的数组。**currentLabel** 属性返回一个字符串，该字符串包含在时间轴上最近遇到的帧标签的名称。

假设创建了一个名为 **robot** 的影片剪辑并已经为其动画的各个状态加上了标签。可以设置一个用于检查 **currentLabel** 属性的条件以访问 **robot** 的当前状态，如下面的代码所示：

```
if (robot.currentLabel == "walking")
{
    // do something
}
```

处理场景

在 Flash 创作环境中，您可以使用场景来区分 SWF 文件播放时将要经过的一系列时间轴。使用 **gotoAndPlay()** 或 **gotoAndStop()** 方法的第二个参数，可以指定要向其发送播放头的场景。所有 FLA 文件开始时都只有初始场景，但您可以创建新的场景。

使用场景并非始终是最佳方法，因为场景有许多缺点。包含多个场景的 Flash 文档可能很难维护，尤其是在存在多个作者的环境中。多个场景也会使带宽效率降低，因为发布过程会将所有场景合并为一个时间轴。这样将使所有场景进行渐进式下载，即使从不会播放这些场景。因此，除非是组织冗长的基于多个时间轴的动画，否则通常不鼓励使用多个场景。

MovieClip 类的 **scenes** 属性返回表示 SWF 文件中所有场景的 **Scene** 对象的数组。**currentScene** 属性返回一个表示当前正在播放的场景的 **Scene** 对象。

Scene 类具有多个提供有关场景信息的属性。**labels** 属性返回表示该场景中帧标签的 **FrameLabel** 对象的数组。**name** 属性以字符串形式返回场景的名称。**numFrames** 属性返回一个表示场景中帧的总数的整数。

使用 ActionScript 创建 MovieClip 对象

在 Flash 中向屏幕中添加内容的一个方法是将资源从库中拖放到舞台上，但不是仅有这种方法。对于复杂项目，经验丰富的开发人员通常更喜欢以编程方式创建影片剪辑。这种方法具有多个优点：代码更易于重用、编译时速度加快，以及仅可在 ActionScript 中进行的更复杂的修改。

ActionScript 3.0 的显示列表 API 简化了动态创建 **MovieClip** 对象的过程。直接实例化 **MovieClip** 实例的功能从向显示列表中添加该实例的过程中分离出来，从而更加灵活、简单，而不会牺牲控制性能。

在 ActionScript 3.0 中，当以编程方式创建影片剪辑（或任何其它显示对象）实例时，只有通过对显示对象容器调用 **addChild()** 或 **addChildAt()** 方法将该实例添加到显示列表中后，才能在屏幕上看到该实例。这允许您创建影片剪辑、设置其属性，甚至可以在向屏幕呈现该影片剪辑之前调用方法。有关使用显示列表的详细信息，请参阅第 232 页的“[处理显示对象容器](#)”。

为 ActionScript 导出库元件

默认情况下，Flash 文档库中的影片剪辑元件实例不能以动态方式创建（即只使用 ActionScript 创建）。这是因为导出供 ActionScript 使用的每个元件都会增加 SWF 文件的大小，而且众所周知，有些元件可能不适合在舞台上使用。因此，为了使元件可以在 ActionScript 中使用，必须指定为 ActionScript 导出该元件。

为 ActionScript 导出元件：

- 1 在“库”面板中选择该元件并打开其“元件属性”对话框。
- 2 必要时激活“高级”设置。
- 3 在“链接”部分中，激活“为 ActionScript 导出”复选框。

这将激活“类”和“基类”字段。

默认情况下，“类”字段会用删除了空格的元件名称进行填充（例如，名为“Tree House”的元件会变为“TreeHouse”）。若要指定该元件对其行为使用自定义类，请在此字段中输入该类的完整名称，包括它所在的包。如果希望在 ActionScript 中创建该元件的实例，但不需要添加任何其它行为，则可以使类名称保持原样。

“基类”字段的值默认为 `flash.display.MovieClip`。如果想让元件扩展另一个自定义类的功能，可以指定该类的名称替代这个值，只要该类扩展 `Sprite`（或 `MovieClip`）类即可。

- 4 按“确定”按钮以保存所做的更改。

此时，如果 Flash 无法找到包含指定类的定义的外部 ActionScript 文件（例如，如果不需要为元件添加其它行为），则会显示警告：

无法在类路径中找到对此类的定义，因此将在导出时自动在 SWF 文件中生成相应的定义。

如果库元件不需要超出 `MovieClip` 类功能的独特功能，则可以忽略此警告消息。

如果没有为元件提供类，Flash 将为元件创建一个等同于下面所示类的类：

```
package
{
    import flash.display.MovieClip;

    public class ExampleMovieClip extends MovieClip
    {
        public function ExampleMovieClip()
        {
        }
    }
}
```

如果想要向元件中添加额外的 ActionScript 功能，请向下面的代码结构中添加相应的属性和方法。例如，假如有一个包含 50 像素宽和 50 像素高的圆形的影片剪辑元件，并用名为 `Circle` 的类指定为 ActionScript 导出该元件。以下代码在放入 `Circle.as` 文件后将扩展 `MovieClip` 类，同时为此元件提供额外的方法 `getArea()` 和 `getCircumference()`：

```
package
{
    import flash.display.MovieClip;

    public class Circle extends MovieClip
    {
        public function Circle()
        {
        }

        public function getArea():Number
        {
            // The formula is Pi times the radius squared.
            return Math.PI * Math.pow((width / 2), 2);
        }

        public function getCircumference():Number
        {
            // The formula is Pi times the diameter.
            return Math.PI * width;
        }
    }
}
```

放置在 Flash 文档第 1 帧的关键帧上的以下代码将创建该元件的一个实例，并在屏幕上显示该实例：

```
var c:Circle = new Circle();
addChild(c);
trace(c.width);
trace(c.height);
trace(c.getArea());
trace(c.getCircumference());
```

此代码演示了基于 ActionScript 的实例化可作为将单个资源拖放到舞台上的替代方法。它所创建的圆形具有影片剪辑的所有属性，同时还具有 Circle 类中定义的自定义方法。这是一个非常简单的示例 — 您的库元件可在其类中指定任意数目的属性和方法。

基于 ActionScript 的实例化功能强大，因为允许动态创建大量实例，而如果采用手动方式来创建将是一项繁重的任务。同时还很灵活，因为您可以在创建每个实例时自定义该实例的属性。您可以通过使用循环动态创建多个 Circle 实例来体会上述优点。在 Flash 文档库中存在上述 Circle 元件和类的情况下，将下面的代码放在第 1 帧的关键帧上：

```
import flash.geom.ColorTransform;

var totalCircles:uint = 10;
var i:uint;
for (i = 0; i < totalCircles; i++)
{
    // Create a new Circle instance.
    var c:Circle = new Circle();
    // Place the new Circle at an x coordinate that will space the circles
    // evenly across the Stage.
    c.x = (stage.stageWidth / totalCircles) * i;
    // Place the Circle instance at the vertical center of the Stage.
    c.y = stage.stageHeight / 2;
    // Change the Circle instance to a random color
    c.transform.colorTransform = getRandomColor();
    // Add the Circle instance to the current timeline.
    addChild(c);
}

function getRandomColor():ColorTransform
{
    // Generate random values for the red, green, and blue color channels.
    var red:Number = (Math.random() * 512) - 255;
    var green:Number = (Math.random() * 512) - 255;
    var blue:Number = (Math.random() * 512) - 255;

    // Create and return a ColorTransform object with the random colors.
    return new ColorTransform(1, 1, 1, 1, red, green, blue, 0);
}
```

此代码演示了如何使用代码快速创建和自定义元件的多个实例。每个实例都根据循环内的当前计数进行定位，并且每个实例都通过设置 transform 属性（Circle 通过扩展 MovieClip 类而继承该属性）获得了一种随机颜色。

加载外部 SWF 文件

在 ActionScript 3.0 中，SWF 文件是使用 Loader 类来加载的。若要加载外部 SWF 文件，ActionScript 需要执行以下 4 个操作：

- 1 用文件的 URL 创建一个新的 URLRequest 对象。
- 2 创建一个新的 Loader 对象。
- 3 调用 Loader 对象的 load() 方法，并以参数形式传递 URLRequest 实例。
- 4 对显示对象容器（如 Flash 文档的主时间轴）调用 addChild() 方法，将 Loader 实例添加到显示列表中。

最后，代码如下所示：

```
var request:URLRequest = new URLRequest("http://www.[yourdomain].com/externalSwf.swf");
var loader:Loader = new Loader();
loader.load(request);
addChild(loader);
```

通过指定图像文件的 URL 而不是 SWF 文件的 URL，可以使用上述同样的代码加载外部图像文件，如 JPEG、GIF 或 PNG 图像。SWF 文件不同于图像文件，可能包含 ActionScript。因此，虽然加载 SWF 文件的过程可能与加载图像的过程完全相同，但如果 Flash Player 或 AIR 在播放 SWF，并且您计划使用 ActionScript 以某种方式与外部 SWF 文件通信，则在加载该外部 SWF 文件时，执行加载的 SWF 文件和要加载的 SWF 文件必须位于同一个安全沙箱中。另外，如果外部 SWF 文件包含了与执行加载的 SWF 文件中的类共享同一命名空间的类，可能需要为被加载的 SWF 文件创建新的应用程序域才能避免命名空间冲突。有关安全和应用程序域注意事项的详细信息，请参阅第 547 页的“[使用 ApplicationDomain 类](#)”和第 598 页的“[加载 SWF 文件和图像](#)”。

当成功加载外部 SWF 文件后，可通过 Loader.content 属性访问该文件。如果该外部 SWF 文件是针对 ActionScript 3.0 发布的，则加载的文件将为影片剪辑或 sprite，具体取决于所扩展的类。

加载早期 SWF 文件的注意事项

如果已使用早期版本的 ActionScript 发布了外部 SWF 文件，则需要考虑一些重要的限制条件。与在 AVM2 (ActionScript Virtual Machine 2) 中运行的 ActionScript 3.0 SWF 文件不同，针对 ActionScript 1.0 或 2.0 发布的 SWF 文件在 AVM1 (ActionScript Virtual Machine 1) 中运行。

成功加载 AVM1 SWF 文件后，已加载的对象（Loader.content 属性）将是 AVM1Movie 对象。AVM1Movie 实例不同于 MovieClip 实例。而是显示对象，但不同于影片剪辑，它不包括与时间轴相关的方法或属性。父级 AVM2 SWF 文件将不能访问已加载的 AVM1Movie 对象的属性、方法或对象。

对 AVM2 SWF 文件加载的 AVM1 SWF 文件还有其它限制。有关详细信息，请参阅《ActionScript 3.0 语言和组件参考》中的 AVM1Movie 类列表。

示例：RuntimeAssetsExplorer

“为 ActionScript 导出”功能非常适合用于在多个项目间使用库的情况。如果 Flash Player 或 AIR 执行 SWF 文件，则任何 SWF 文件只要与加载它的 SWF 处于同一安全沙箱中就可以使用已导出到 ActionScript 的元件。这样，单个 Flash 文档可生成仅用于保存图形资源的 SWF 文件。此技术对大型项目特别有用，在这样的项目中，处理可视资源的设计人员可与创建“包装”SWF 文件的开发人员同时工作，然后在运行时加载图形资源 SWF 文件。您可以使用此方法维护一系列图形资源与编程开发进度无关的版本文件。

RuntimeAssetsExplorer 应用程序加载属于 RuntimeAsset 子类的任何 SWF 文件，并允许您浏览该 SWF 文件的可用资源。示例说明了以下过程：

- 使用 Loader.load() 加载外部 SWF 文件
- 动态创建为 ActionScript 导出的库元件
- 使用 ActionScript 控制 MovieClip 回放

开始之前，注意每个要在 Flash Player 中运行的 SWF 文件必须位于同一个安全沙箱中。有关详细信息，请参阅第 588 页的“[安全沙箱](#)”。

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。

RuntimeAssetsExplorer 应用程序文件位于文件夹 Samples/RuntimeAssetsExplorer 中。该应用程序包含以下文件：

文件	描述
RuntimeAssetsExample.mxml 或 RuntimeAssetsExample.fla	适用于 Flex (MXML) 或 Flash (FLA) 的应用程序的用户界面。
RuntimeAssetsExample.as	Flash (FLA) 应用程序的文档类。
GeometricAssets.as	用于实现 RuntimeAsset 接口的示例类。
GeometricAssets.fla	链接到 GeometricAssets 类的 FLA 文件（该类是 FLA 的文档类）包含为 ActionScript 导出的元件。

文件	描述
com/example/programmingas3/runtimeassetexplorer/RuntimeLibrary.as	用于定义将加载到浏览器容器中的所有运行时资源 SWF 文件所需方法的接口。
com/example/programmingas3/runtimeassetexplorer/AnimatingBox.as	形状为旋转方框的库元件的类。
com/example/programmingas3/runtimeassetexplorer/AnimatingStar.as	形状为旋转星形的库元件的类。

建立运行时库界面

若要使浏览器与 SWF 库正确交互，运行时资源库的结构就必须具有一定程式。我们将通过创建一个接口来完成此过程（接口类似于类，因为接口也是用于区分预期结构的方法的蓝图；同时接口又与类不同，因为接口不包含方法体）。接口提供了在运行时库和浏览器之间相互通信的方法。加载到浏览器中的运行时资源的每个 SWF 都将实现此接口。有关接口和如何使用接口的详细信息，请参阅第 87 页的“[接口](#)”。

RuntimeLibrary 接口非常简单 — 我们只需要一个函数，这个函数能够为浏览器提供类路径的数组以导出元件并使元件在运行时库中可用。为此，该接口具有单个方法：`getAssets()`。

```
package com.example.programmingas3.runtimeassetexplorer
{
    public interface RuntimeLibrary
    {
        function getAssets():Array;
    }
}
```

创建资源库 SWF 文件

通过定义 **RuntimeLibrary** 接口，可以创建能够加载到另一个 SWF 文件中的多个资源库 SWF 文件。制作资源的单个 SWF 库包括四个任务：

- 为资源库 SWF 文件创建一个类
- 为库中包含的单个资源创建类
- 创建实际图形资源
- 将图形元素与类关联并发布库 SWF

创建一个类以实现 **RuntimeLibrary** 接口

下一步，我们将创建要实现 **RuntimeLibrary** 接口的 **GeometricAssets** 类。这个类将成为 FLA 的文档类。此类的代码与 **RuntimeLibrary** 接口非常相似，不同之处在于，在类定义中，`getAssets()` 方法有方法体。

```
package
{
    import flash.display.Sprite;
    import com.example.programmingas3.runtimeassetexplorer.RuntimeLibrary;

    public class GeometricAssets extends Sprite implements RuntimeLibrary
    {
        public function GeometricAssets() {

        }

        public function getAssets():Array {
            return [ "com.example.programmingas3.runtimeassetexplorer.AnimatingBox",
                    "com.example.programmingas3.runtimeassetexplorer.AnimatingStar" ];
        }
    }
}
```

如果要创建第二个运行时库，可以另外创建一个基于另一个类（例如 AnimationAssets）的 FLA，该类可提供自带的 getAssets() 实现。

为每个 MovieClip 资源创建类

对于本示例，我们只扩展 MovieClip 类而不为自定义资源添加任何功能。以下 AnimatingStar 的代码类似于 AnimatingBox 的代码：

```
package com.example.programmingas3.runtimeassetexplorer
{
    import flash.display.MovieClip;

    public class AnimatingStar extends MovieClip
    {
        public function AnimatingStar()
        {
        }
    }
}
```

发布库

现在将基于 MovieClip 的资源连接到新类，方法是创建一个新的 FLA 并在“属性”检查器的“文档类”字段中输入 GeometricAssets。为实现本示例的目的，我们将创建两个使用时间轴补间的非常简单的形状，其中一个形状顺时针旋转超过 360 帧。animatingBox 和 animatingStar 元件都设为“为 ActionScript 导出”，并将“类”字段设置为 getAssets() 实现中指定的对应类路径。保留 flash.display.MovieClip 的默认基类，因为我们希望对标准 MovieClip 方法进行子分类。

在设置了元件的导出设置后，可发布 FLA。您现在便拥有了第一个运行时库。该 SWF 文件可以加载到另一个 AVM2 SWF 文件中，且 AnimatingBox 和 AnimatingStar 元件可用于新的 SWF 文件。

将库加载到另一个 SWF 文件中

要处理的最后一个功能性部分是资源浏览器的用户界面。在本示例中，运行时库的路径硬编码为一个名为 ASSETS_PATH 的变量。或者，您也可以使用 FileReference 类，例如，用来创建在硬盘驱动器上浏览特定 SWF 文件的接口。

成功加载运行时库后，Flash Player 会调用 runtimeAssetsLoadComplete() 方法：

```
private function runtimeAssetsLoadComplete(event:Event):void
{
    var rl:* = event.target.content;
    var assetList:Array = rl.getAssets();
    populateDropdown(assetList);
    stage.frameRate = 60;
}
```

在此方法中，变量 rl 表示已加载的 SWF 文件。代码将调用已加载的 SWF 文件的 getAssets() 方法，获取可用资源的列表，并通过调用 populateDropDown() 方法用这些资源填充具有可用资源列表的 ComboBox 组件。该方法会依次存储每个资源的完整类路径。单击界面上的“添加”按钮即会触发 addAsset() 方法：

```
private function addAsset():void
{
    var className:String = assetNameCbo.selectedItem.data;
    var AssetClass:Class = getDefinitionByName(className) as Class;
    var mc:MovieClip = new AssetClass();
    ...
}
```

此方法获取 ComboBox 中当前所选资源的类路径 (assetNameCbo.selectedItem.data)，并使用 getDefinitionByName() 函数（来自 flash.utils 包）获取对该资源的类的实际引用，以创建该资源的新实例。

第 19 章：使用补间动画

第 256 页的“[对象动画](#)”介绍如何在 ActionScript 中实现脚本动画。

我们在这里介绍另一种创建动画的技术：补间动画。借助这种技术，您可以使用 Adobe® Flash® CS4 Professional 通过在 FLA 文件中以交互方式设置动画来创建移动效果。然后您可以在运行时将该动画用于基于 ActionScript 的动态动画。

Flash CS4 可自动生成实现补间动画的 ActionScript，并供您复制和重复使用。

若要创建补间动画，您必须拥有 Adobe Flash CS4 Professional 的许可证。

补间动画基础知识

ActionScript 中的补间动画简介

补间动画提供了创建动画的简单方法。

补间动画以帧到帧的方式修改显示对象属性（如位置或旋转）。在显示对象移动期间，补间动画还可以通过应用各种滤镜和其他属性来更改显示对象的外观。您可使用 Flash 通过交互方式创建补间动画，这会生成用于补间动画的 ActionScript。在 Flash 中，可使用“将动画复制为 ActionScript 3.0 脚本”命令复制创建补间动画的 ActionScript。随后您便可以在您自己的动态动画中重复使用该 ActionScript，以便在运行时创建移动效果。

有关创建补间动画的信息，请参阅《使用 Flash CS4 Professional》中的“补间动画”一节。

常见补间动画任务

用于实现补间动画的、动态生成的 ActionScript 代码实现以下功能：

- 实例化补间动画的动画对象
- 设置补间动画的持续时间
- 添加补间动画的属性
- 将滤镜添加到补间动画
- 将补间动画与其一个或多个显示对象关联

重要术语和概念

下面是一个您将在本章中遇到的重要术语：

- 补间动画：一种构造，它生成显示对象在不同时间不同状态下的中间帧；提供使第一个状态平滑过渡到第二个状态的外观。用于在舞台上移动显示对象，并使显示对象随时间而增大、缩小、旋转、淡化或更改颜色。


复制补间动画脚本

补间生成的中间帧可显示处于时间轴上两个不同帧中的不同状态下的显示对象。它创建的外观可使第一个帧中的图像平滑过渡到第二个帧中的图像。在补间动画中，外观的更改通常涉及显示对象位置的更改，因而可创建移动效果。除了调整显示对象的位置之外，补间动画还可以对显示对象进行旋转、倾斜、调整大小或应用滤镜。

在 Flash 中，通过沿时间轴在两个关键帧之间移动显示对象来创建补间动画。Flash 可自动生成描述补间的 ActionScript 代码，您可以复制这些代码并将其保存在文件中。有关创建补间动画的信息，请参阅《使用 Flash》中的“补间动画”一节。

您可以通过两种方式访问 Flash 中的“将动画复制为 ActionScript 3.0 脚本”命令。第一种方式是通过舞台上的补间上下文菜单：

- 1 在舞台上选择补间动画。
- 2 右键单击 (Windows) 或按住 Control 单击 (Macintosh)。
- 3 选择“将动画复制为 ActionScript 3.0 脚本 ...”



第二种方式是直接从 Flash“编辑”菜单选择该命令：

- 1 在舞台上选择补间动画。
- 2 选择“编辑”>“时间轴”>“将动画复制为 ActionScript 3.0 脚本”。



在复制了脚本后，将脚本粘贴到文件中并保存。

在创建某个补间动画并复制和保存脚本后，您可以按原样重复使用该脚本，也可以在您自己的基于 ActionScript 的动态动画中修改该脚本。

合并补间动画脚本

您从 Flash 复制的 ActionScript 代码中的标头会列出支持补间动画所需的所有模块。

补间动画类

基本类为 fl.motion 包中的 AnimatorFactory、MotionBase 和 Motion 类。根据补间动画所操作的属性，您可能还需要其它类。例如，如果补间动画对显示对象进行变形或旋转，则需要导入相应的 flash.geom 类。如果补间动画应用了滤镜，则需导入 flash.filter 类。在 ActionScript 中，补间动画是 Motion 类的实例。Motion 类存储可应用于可视对象的关键帧动画序列。动画数据包括位置、缩放、旋转、倾斜、颜色、滤镜和缓动。

下面的 ActionScript 是从某个补间动画复制而来的，创建该补间动画是为了对一个实例名称为 Symbol1_2 的显示对象进行动画处理。这段代码为一个名为 __motion_Symbol1_2 的 MotionBase 对象声明了一个变量。MotionBase 类是 Motion 类的父级。

```
var __motion_Symbol1_2:MotionBase;
```

随后脚本创建 Motion 对象：

```
__motion_Symbol1_2 = new Motion();
```

Motion 对象名称

在前例中，Flash 自动为 Motion 对象生成名称 __motion_Symbol1_2。它将前缀 __motion_ 附加到显示对象名称。因而在 Flash 创作工具中，自动生成的名称基于补间动画的目标对象的实例名称。Motion 对象的 duration 属性指示补间动画中的总帧数：

```
__motion_Symbol1_2.duration = 200;
```

当您在自己的动画中重复使用这样的 ActionScript 时，您可以保留 Flash 为补间自动生成的名称。也可以替换为另一个名称。默认情况下，如果您要复制的补间动画所属的显示对象实例还没有实例名称，则 Flash 会自动对该实例进行命名。如果您更改了补间名称，请确保在整个脚本中都更改该名称。或者，您可以在 Flash 中将您选择的名称分配给补间动画的目标对象。然后创建补间动画并复制脚本。无论您使用哪种命名方式，都要确保您的 ActionScript 代码中的每个 Motion 对象都有唯一的名称。

描述动画

MotionBase 类的 addPropertyArray() 方法添加用于描述每个补间属性的值数组。

该数组可能包含用于补间动画中每个关键帧的数组项。通常，这些数组中的一些数组包含的项数少于补间动画中的关键帧总数。当数组中的最后一个值对于剩下的帧没有更改时，会发生这种情况。

如果数组参数的长度大于 Motion 对象的 duration 属性，则 addPropertyArray() 会相应地调整 duration 属性的值。它不会为以前添加的属性添加关键帧。会为动画的额外帧保持新添加的关键帧。

Motion 对象的 x 和 y 属性描述补间对象在动画运行时不断更改的位置。如果显示对象的位置发生更改，则这些坐标是在每个关键帧中最可能发生更改的值。您可以使用 addPropertyArray() 方法添加其它动画属性。例如，如果调整补间对象的大小，可以添加 scaleX 和 scaleY 值。如果倾斜补间对象，则添加 skewX 和 skewY 值。如果旋转补间对象，则添加 rotationConcat 属性。

使用 addPropertyArray() 方法可定义以下补间属性：

x	对象的变形点在其父级的坐标空间中的水平位置
y	对象的变形点在其父级的坐标空间中的垂直位置
z	对象的变形点在其父级的坐标空间中的深度 (z 轴) 位置
scaleX	从变形点开始应用的对象的水平缩放比例 (百分比)
scaleY	从变形点开始应用的对象的垂直缩放比例 (百分比)
skewX	从变形点开始应用的对象的水平倾斜角度 (以度为单位)

skewY	从变形点开始应用的对象的垂直倾斜角度（以度为单位）
rotationX	对象相对于其原始方向围绕 x 轴的旋转
rotationY	对象相对于其原始方向围绕 y 轴的旋转
rotationConcat	动画中的对象的旋转（z 轴）值，相对于前一个方向且从变形点开始应用
useRotationConcat	如果设置此属性，则会在 addPropertyArray() 提供动画数据时导致目标对象旋转
blendMode	BlendMode 类值，指定对象的颜色与底层图形的混合
matrix3D	matrix3D 属性（如果对于关键帧存在此属性）；用于 3D 补间；如果使用，则会忽略以前所有的变形属性
rotationZ	对象相对于 3D 父容器从其原始方向开始的 z 轴旋转（以度为单位）；用于 3D 补间（而不是 rotationConcat）

在自动生成的脚本中添加的属性取决于在 Flash 中分配给补间动画的属性。在自定义自己的脚本版本时，您可以添加、删除或修改其中一些属性。

下面的代码向名为 `_motion_Wheel` 的补间动画的属性赋值。在此示例中，补间显示对象在补间动画中的全部 29 个帧中都不更改位置，而是在当前位置上旋转。分配给 `rotationConcat` 数组的多个值对旋转进行了定义。此补间动画的其它属性值无变化。

```

_motion_Wheel = new Motion();
_motion_Wheel.duration = 29;
_motion_Wheel.addPropertyArray("x", [0]);
_motion_Wheel.addPropertyArray("y", [0]);
_motion_Wheel.addPropertyArray("scaleX", [1.00]);
_motion_Wheel.addPropertyArray("scaleY", [1.00]);
_motion_Wheel.addPropertyArray("skewX", [0]);
_motion_Wheel.addPropertyArray("skewY", [0]);
_motion_Wheel.addPropertyArray("rotationConcat",
[
    0,-13.2143,-26.4285,-39.6428,-52.8571,-66.0714,-79.2857,-92.4999,-105.714,
    -118.929,-132.143,-145.357,-158.571,-171.786,-185,-198.214,-211.429,-224.643,
    -237.857,-251.071,-264.286,-277.5,-290.714,-303.929,-317.143,-330.357,
    -343.571,-356.786,-370
]);
_motion_Wheel.addPropertyArray("blendMode", ["normal"]);

```

在下一个示例中，名为 `Leaf_1` 的显示对象会在舞台上移动。该对象的 x 和 y 属性数组为动画的 100 个帧中的每个帧包含不同的值。此外，该对象在舞台上移动时同时在其 z 轴上旋转。`rotationZ` 属性数组中的多个项确定了旋转方式。

```
_motion_Leaf_1 = new MotionBase();
_motion_Leaf_1.duration = 100;
_motion_Symbol1_4.addPropertyArray("y",
    [
        0,5.91999,11.84,17.76,23.68,29.6,35.52,41.44,47.36,53.28,59.2,65.12,71.04,
        76.96,82.88,88.8,94.72,100.64,106.56,112.48,118.4,124.32,130.24,136.16,142.08,
        148,150.455,152.909,155.364,157.818,160.273,162.727,165.182,167.636,170.091,
        172.545,175,177.455,179.909,182.364,184.818,187.273,189.727,192.182,194.636,
        197.091,199.545,202,207.433,212.865,218.298,223.73,229.163,234.596,240.028,
        245.461,250.893,256.326,261.759,267.191,272.624,278.057,283.489,
        288.922,294.354,299.787,305.22,310.652,316.085,321.517,326.95,330.475,334,
        337.525,341.05,344.575,348.1,351.625,355.15,358.675,362.2,365.725,369.25,
        372.775,376.3,379.825,383.35,386.875,390.4,393.925,397.45,400.975,404.5,
        407.5,410.5,413.5,416.5,419.5,422.5,425.5
    ]
);
_motion_Symbol1_4.addPropertyArray("scaleX", [1.00]);
_motion_Symbol1_4.addPropertyArray("scaleY", [1.00]);
_motion_Symbol1_4.addPropertyArray("skewX", [0]);
_motion_Symbol1_4.addPropertyArray("skewY", [0]);
_motion_Symbol1_4.addPropertyArray("z",
    [
        0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,
        0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,
        0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
    ]
);
_motion_Symbol1_4.addPropertyArray("rotationX", [64.0361]);
_motion_Symbol1_4.addPropertyArray("rotationY", [41.9578]);
_motion_Symbol1_4.addPropertyArray("rotationZ",
    [
        -18.0336,-17.5536,-17.0736,-16.5936,-16.1136,-15.6336,-15.1536,-14.6736,
        -14.1936,-13.7136,-13.2336,-12.7536,-12.2736,-11.7936,-11.3136,-10.8336,
        -10.3536,-9.8736,-9.3936,-8.9136,-8.4336,-7.9536,-7.4736,-6.9936,-6.5136,
        -6.0336,-7.21542,-8.39723,-9.57905,-10.7609,-11.9427,-13.1245,-14.3063,
        -15.4881,-16.67,-17.8518,-19.0336,-20.2154,-21.3972,-22.5791,-23.7609,
        -24.9427,-26.1245,-27.3063,-28.4881,-29.67,-30.8518,-32.0336,-31.0771,
        -30.1206,-29.164,-28.2075,-27.251,-26.2945,-25.338,-24.3814,-23.4249,
        -22.4684,-21.5119,-20.5553,-19.5988,-18.6423,-17.6858,-16.7293,-15.7727
        -14.8162,-13.8597,-12.9032,-11.9466,-10.9901,-10.0336,-10.9427,-11.8518,
        -12.7609,-13.67,-14.5791,-15.4881,-16.3972,-17.3063,-18.2154,-19.1245,
        -20.0336,-20.9427,-21.8518,-22.7609,-23.67,-24.5791,-25.4881,-26.3972,
        -27.3063,-28.2154,-29.1245,-30.0336,-28.3193,-26.605,-24.8907,-23.1765,
        -21.4622,-19.7479,-18.0336
    ]
);
_motion_Symbol1_4.addPropertyArray("blendMode", ["normal"]);
```

添加滤镜

如果补间动画的目标对象包含滤镜，则这些滤镜是使用 Motion 类的 initFilters() 和 addFilterPropertyArray() 方法添加的。

初始化滤镜数组

initFilters() 方法将初始化滤镜。该方法的第一个参数是应用于显示对象的所有滤镜的完全限定类名称的数组。此滤镜名称数组是从 Flash 中的补间动画滤镜列表生成的。在您的脚本副本中，您可以在此数组中删除或添加 flash.filters 包中的任何滤镜。下面的调用初始化目标显示对象的滤镜列表。该调用应用 DropShadowFilter、GlowFilter 和 BevelFilter，并将列表复制到 Motion 对象中的每个关键帧。

```
_motion_Box.initFilters(["flash.filters.DropShadowFilter", "flash.filters.GlowFilter",
"flash.filters.BevelFilter"], [0, 0, 0]);
```

添加滤镜

addFilterPropertyArray() 方法使用以下参数描述已初始化滤镜的属性：

- 1 该方法的第一个参数通过索引标识滤镜。该索引指示滤镜名称在滤镜类名称数组中的位置，该数组是在前一个 initFilters() 调用中传递的。
- 2 该方法的第二个参数是在每个关键帧中为该滤镜存储的滤镜属性。
- 3 该方法的第三个参数是指定滤镜属性的值。

在进行了前一个 initFilters() 调用后，随后的 addFilterPropertyArray() 调用将值 5 分配给 DropShadowFilter 的 blurX 和 blurY 属性。DropShadowFilter 是已初始化的滤镜数组中的第一个（索引为 0）项：

```
_motion_Box.addFilterPropertyArray(0, "blurX", [5]);
_motion_Box.addFilterPropertyArray(0, "blurY", [5]);
```

接下来的三个调用向已初始化滤镜数组中的第二个项（索引为 1） GlowFilter 的 quality、alpha 和 color 属性赋值。

```
_motion_Box.addFilterPropertyArray(1, "quality", [BitmapFilterQuality.LOW]);
_motion_Box.addFilterPropertyArray(1, "alpha", [1.00]);
_motion_Box.addFilterPropertyArray(1, "color", [0xff0000]);
```

接下来的四个调用向已初始化滤镜数组的第三个项（索引为 2） BevelFilter 的 shadowAlpha、shadowColor、highlightAlpha 和 highlightColor 赋值。

```
_motion_Box.addFilterPropertyArray(2, "shadowAlpha", [1.00]);
_motion_Box.addFilterPropertyArray(2, "shadowColor", [0x000000]);
_motion_Box.addFilterPropertyArray(2, "highlightAlpha", [1.00]);
_motion_Box.addFilterPropertyArray(2, "highlightColor", [0xffffffff]);
```

使用 ColorMatrixFilter 调整颜色

在初始化 ColorMatrixFilter 之后，您可以设置相应的 AdjustColor 属性来调整补间显示对象的亮度、对比度、饱和度和色相。通常，您在 Flash 中应用 AdjustColor 滤镜，然后可以在您的 ActionScript 副本中调整该滤镜。下面的示例在显示对象移动期间转换该对象的色相和饱和度。

```

__motion_Leaf_1.initFilters(["flash.filters.ColorMatrix"], [0], -1, -1);
__motion_Leaf_1.addFilterPropertyArray(0, "adjustColorBrightness", [0], -1, -1);
__motion_Leaf_1.addFilterPropertyArray(0, "adjustColorContrast", [0], -1, -1);
__motion_Leaf_1.addFilterPropertyArray(0, "adjustColorSaturation",
[
    [
        0,-0.589039,1.17808,-1.76712,-2.35616,-2.9452,-3.53424,-4.12328,
        -4.71232,-5.30136,-5.89041, 6.47945,-7.06849,-7.65753,-8.24657,
        -8.83561,-9.42465,-10.0137,-10.6027,-11.1918,11.7808,-12.3699,
        -12.9589,-13.5479,-14.137,-14.726,-15.3151,-15.9041,-16.4931,
        17.0822,-17.6712,-18.2603,-18.8493,-19.4383,-20.0274,-20.6164,
        -21.2055,-21.7945,22.3836,-22.9726,-23.5616,-24.1507,-24.7397,
        -25.3288,-25.9178,-26.5068,-27.0959,27.6849,-28.274,-28.863,-29.452,
        -30.0411,-30.6301,-31.2192,-31.8082,-32.3973,32.9863,-33.5753,
        -34.1644,-34.7534,-35.3425,-35.9315,-36.5205,-37.1096,-37.6986,
        38.2877,-38.8767,-39.4657,-40.0548,-40.6438,-41.2329,-41.8219,
        -42.411,-43
    ],
    -1, -1);
__motion_Leaf_1.addFilterPropertyArray(0, "adjustColorHue",
[
    0,0.677418,1.35484,2.03226,2.70967,3.38709,4.06451,4.74193,5.41935,
    6.09677,6.77419,7.45161,8.12903,8.80645,9.48387,10.1613,10.8387,11.5161,
    12.1935,12.871,13.5484,14.2258,14.9032,15.5806,16.2581,16.9355,17.6129,
    18.2903,18.9677,19.6452,20.3226,21,22.4286,23.8571,25.2857,26.7143,28.1429,
    29.5714,31,32.4286,33.8571,35.2857,36.7143,38.1429,39.5714,41,42.4286,43.8571,
    45.2857,46.7143,48.1429,49.5714,51,54,57,60,63,66,69,72,75,78,81,84,87,
    90,93,96,99,102,105,108,111,114
],
    -1, -1);

```

将补间动画与其显示对象关联

最后一项任务是将补间动画与它所操作的一个或多个显示对象关联起来。

AnimatorFactory 类管理补间动画与其目标显示对象之间的关联。 **AnimatorFactory** 构造函数的参数为 **Motion** 对象：

```
var __animFactory_Wheel:AnimatorFactory = new AnimatorFactory(__motion_Wheel);
```

使用 **AnimatorFactory** 类的 **addTarget()** 方法可将目标显示对象与其补间动画关联起来。从 Flash 复制的 ActionScript 注释掉了 **addTarget()** 代码行，并且未指定实例名称：

```
// __animFactory_Wheel.addTarget(<instance name goes here>, 0);
```

在您的副本中，指定要与补间动画关联的显示对象。在下面的示例中，目标指定为 **greenWheel** 和 **redWheel**：

```
__animFactory_Wheel.AnimatorFactory.addTarget(greenWheel, 0);
__animFactory_Wheel.AnimationFactory.addTarget(redWheel, 0);
```

您可以使用多个 **addTarget()** 调用，将多个显示对象与同一个补间动画关联起来。

第 20 章：使用反向运动

反向运动 (IK) 是一种用于创建逼真运动的重要方法。

使用 IK 可以在一系列连接的部分（称为 IK 骨架）中创建协调运动，以便各部分以逼真的方式一起移动。骨架的各部分是其骨骼和连接。如果给定骨架的终点，IK 便可计算出达到该终点所需的连接的角度。

自己手动计算这些角度难度很大。此功能的优点在于您可以使用 **Adobe® Flash® CS4 Professional** 以交互方式创建骨架。然后可以使用 ActionScript 对骨架进行动画处理。Flash 创作工具中附带的 IK 引擎可进行计算，以描述骨架的移动。您可以在 ActionScript 代码中使用某些参数限制移动。

若要创建反向运动骨架，您必须拥有 **Adobe Flash CS4 Professional** 的许可证。

反向运动的基础知识

IK 简介

使用反向运动 (IK) 可以将不同的部分链接起来，使它们以逼真的方式进行相对移动，从而创建逼真的动画效果。

例如，使用 IK 时，您可以根据需要移动某条腿的连接将那条腿移动到特定的位置，从而获得所需的姿势。IK 使用一种骨骼框架，这些骨骼采用称为“IK 骨架”的结构连接在一起。**flik** 包可帮助您创建模仿自然运动的动画。使用该包可以对多个 IK 骨架进行无缝动画处理，而不必了解很多 IK 算法所依赖的物理知识。

使用 Flash 创建 IK 骨架及其辅助骨骼和连接。然后您便可以在运行时访问 IK 类以对它们进行动画处理。

有关如何创建 IK 骨架的详细介绍，请参阅《使用 Flash CS4 Professional》中的“使用反向运动”一节。

常见 IK 任务

在运行时启动和控制 IK 骨架移动的 ActionScript 代码通常完成以下功能：

- 为运动所涉及的骨架、骨骼和连接声明变量
- 检索骨架、骨骼和连接实例
- 实例化 IKMover 对象
- 设置对移动的限制
- 将骨架移动到目标点

重要术语和概念


以下参考列表包含将会在本章中遇到的重要术语：

- 骨架：一种由骨架和连接组成的运动链，用于计算机动画以模拟逼真的运动
- 骨骼：骨架中不能弯曲的段，类似于动物骨架中的骨骼
- 反向运动 (IK)：确定连接的灵活对象（称为运动链或骨架）的参数的过程
- 连接：两个骨骼相接的位置，用于实现骨骼的移动；类似于动物的关节

IK 骨架动画处理概述


在创建 IK 骨架后，可使用 fl.ik 类在运行时限制骨架的移动、跟踪其事件并对其进行动画处理。

下图演示了一个名为 Wheel 的影片剪辑。轴是一个名为 Axle 的 IKArmature 实例。IKMover 类会在轮子旋转时同步移动骨架。骨架中的 IKBone (ikBone2) 在其尾部连接处附加到轮子。



A. Wheel B. Axe C. ikBone2

在运行时，轮子结合 __motion_Wheel 补间动画（在“使用补间动画”一章中的第 349 页的“[描述动画](#)”中进行了介绍）进行旋转。由一个 IKMover 对象启动和控制轴的移动。下图演示了附加到旋转轮上的轴骨架在旋转中的不同帧上的两个快照。



在运行时，下面的 ActionScript 可完成以下功能：

- 获取有关骨架及其组件的信息
- 实例化 IKMover 对象
- 结合轮子的旋转移动轴

```

import fl.ik.*;

var tree:IKAarmature = IKManager.getArmatureByName("Axe");
var bone:IKBone = tree.getBoneByName("ikBone2");
var endEffector:IKJoint = bone.tailJoint;
var pos:Point = endEffector.position;

var ik:IKMover = new IKMover(endEffector, pos);
ik.limitByDistance = true;
ik.distanceLimit = 0.1;
ik.limitByIteration = true;
ik.iterationLimit = 10;

Wheel.addEventListener(Event.ENTER_FRAME, frameFunc);

function frameFunc(event:Event)
{
    if (Wheel != null)
    {
        var mat:Matrix = Wheel.transform.matrix;
        var pt = new Point(90, 0);
        pt = mat.transformPoint(pt);

        ik.moveTo(pt);
    }
}

```

用于移动轴的 IK 类为：

- IKAarmature：描述骨架（由骨骼和连接组成的树结构）；必须使用 Flash 创建
- IKManager：文档中所有 IK 骨架的容器类；必须使用 Flash 创建
- IKBone：IK 骨架的一段
- IKJoint：两个 IK 骨骼之间的连接
- IKMover：启动和控制骨架的 IK 移动

有关这些类的完整而详细的说明，请参阅 [ik 包](#)。

获取有关 IK 骨架的信息

首先，为组成要移动的各部分的骨架、骨骼和连接声明变量。

下面的代码使用 IKManager 类的 getArmatureByName() 方法将 Axe 骨架的值分配给 IKAarmature 变量 tree。Axe 骨架是先前使用 Flash 创建的。

```
var tree:IKAarmature = IKManager.getArmatureByName("Axe");
```

同样，下面的代码使用 IKAarmature 类的 getBoneByName() 方法将 ikBone2 骨骼的值分配给 IKBone 变量。

```
var bone:IKBone = tree.getBoneByName("ikBone2");
```

ikBone2 骨骼的尾部连接是附加到旋转轮的骨架部分。

下行代码声明变量 endEffector 并将 ikBone2 骨骼的 tailjoint 属性分配给该变量：

```
var endEffector:IKJoint = bone.tailJoint;
```

变量 pos 是用于存储 endEffector 连接的当前位置的点。

```
var pos:Point = endEffector.position;
```

在此示例中，`pos` 是处于轴尾部（轴在此处与轮子连接）的连接的位置。此变量的原始值是通过 `IKJoint` 的 `position` 属性获取的。

实例化 `IKMover` 并限制其移动

由 `IKMover` 类的实例移动轴。

下行代码实例化 `IKMover` 对象 `ik`，将要移动的元素和移动的起始点传递给该对象的构造函数：

```
var ik:IKMover = new IKMover(endEffector, pos);
```

使用 `IKMover` 类的属性可以限制骨架的移动。可以基于移动的距离、迭代和时间来限制移动。

以下几对属性可强制执行这些限制。这几对属性包括一个用于指示是否限制移动的 Boolean 和一个用于指定限制的数字：

<code>limitByDistance:Boolean</code>	<code>distanceLimit:int</code>	设置 IK 引擎对每次迭代移动的最大距离（以像素为单位）。
<code>limitByIteration:Boolean</code>	<code>iterationLimit:int</code>	设置 IK 引擎对每个移动执行的最大迭代次数。
<code>limitByTime:Boolean</code>	<code>timeLimit:int</code>	设置分配给 IK 引擎用于执行移动的最长时间（以毫秒为单位）。

将相应的 Boolean 属性设置为 `true` 可强制执行限制。默认情况下，所有 Booleans 都设置为 `false`，因此除非您显式设置这些属性，否则移动不会受限制。如果您将限制设置为某个值，却没有设置其对应的 Boolean，则会忽略该限制。在这种情况下，IK 引擎继续移动对象，直至达到其它限制或 `IKMover` 的目标位置。

在下面的示例中，骨架移动的最大距离设置为每次迭代 0.1 个像素。每个移动的最大迭代次数设置为 10。

```
ik.limitByDistance = true;
ik.distanceLimit = 0.1;
ik.limitByIteration = true;
ik.iterationLimit = 10;
```

移动 IK 骨架

`IKMover` 可在轮子的事件侦听器中移动轴。在轮子的每个 `enterFrame` 事件中，都会计算骨架的新目标位置。`IKMover` 使用其 `moveTo()` 方法将尾部连接移动到其目标位置，或是在通过其 `limitByDistance`、`limitByIteration` 和 `limitByTime` 属性设置的限制下移动到尽可能远的位置。

```
Wheel.addEventListener(Event.ENTER_FRAME, frameFunc);

function frameFunc(event:Event)
{
    if (Wheel != null)
    {
        var mat:Matrix = Wheel.transform.matrix;
        var pt = new Point(90,0);
        pt = mat.transformPoint(pt);

        ik.moveTo(pt);
    }
}
```

使用 IK 事件

使用 `IKEvent` 类可以创建包含有关 IK 事件的信息的事件对象。`IKEvent` 信息描述由于超过指定时间、距离或迭代限制而终止的运动。

下面的代码演示用于跟踪时间限制事件的一个事件侦听器和处理函数。此事件处理函数在超过 `IKMover` 的时间限制时，报告所引发的事件的时间、距离、迭代计数和连接属性。

```
var ikmover:IKMover = new IKMover(endjoint, pos);
ikMover.limitByTime = true;
ikMover.timeLimit = 1000;

ikmover.addEventListener(IKEVENT.TIME_LIMIT, timeLimitFunction);

function timeLimitFunction(evt:IKEVENT):void
{
    trace("timeLimit hit");
    trace("time is " + evt.time);
    trace("distance is " + evt.distance);
    trace("iterationCount is " + evt.iterationCount);
    trace("IKJoint is " + evt.joint.name);
}
```

第 21 章：处理文本

在 Adobe® Flash® Player 或 Adobe® AIR™ 中，若要在屏幕上显示文本，可以使用 `TextField` 类的实例或使用 Flash 文本引擎类。这些类可用于执行文本的创建、显示和格式设置。

您可以为文本字段确定具体内容，或者指定文本来源，然后设置该文本的外观。还可以在用户输入文本或单击超文本链接时响应用户事件。

处理文本的基础知识

处理文本简介

`TextField` 类和 Flash 文本引擎类都可用于在 Flash Player 和 AIR 中显示和管理文本。

可以使用 `TextField` 类来创建用于显示和输入的文本对象。`TextField` 类是 Flash 和 Adobe Flex 中提供的其它基于文本的组件（如 `TextArea` 和 `TextInput`）的基础。可以使用 `TextFormat` 类来设置 `TextField` 对象的字符和段落格式，可以使用 `Textfield.styleSheet` 属性和 `StyleSheet` 类来应用层叠样式表 (CSS)。可将 HTML 格式的文本直接分配给文本字段，HTML 格式的文本可包含嵌入的媒体（影片剪辑、SWF 文件、GIF 文件、PNG 文件和 JPEG 文件）。

从 Flash Player 10 和 Adobe AIR 1.5 开始提供的 Flash 文本引擎为文本度量、格式设置和双向文本的复杂控制提供底层支持。它还提供改进的文本流和增强的语言支持。虽然可以使用 Flash 文本引擎来创建和管理文本元素，但它主要是用作创建文本处理组件的基础，且需要更多编程专业技能。

处理文本的常见任务

下面是涉及 `TextField` 类的常见任务：

- 修改文本字段内容
- 在文本字段中使用 HTML
- 在文本字段中使用图像
- 选择文本并处理用户选择的文本
- 捕获文本输入
- 限制文本输入
- 对文本应用格式设置和 CSS 样式
- 使用清晰度、粗细和消除锯齿功能来微调文本显示
- 在 ActionScript 中访问和处理静态文本字段

下面是涉及 Flash 文本引擎类的常见任务：

- 创建和显示文本
- 处理事件
- 设置文本格式
- 使用字体
- 控制文本

重要概念和术语

以下参考列表包含您将遇到的重要术语：

- 层叠样式表：对以 XML（或 HTML）格式构成的内容指定样式和格式设置的标准语法。
- 设备字体：安装在用户计算机上的一种字体。
- 动态文本字段：可以由 ActionScript 更改内容而不能由用户输入内容的文本字段。
- 嵌入字体：将字符轮廓数据存储在应用程序 SWF 文件中的一种字体。
- HTML 文本：使用 ActionScript 输入到文本字段中的文本内容，包括 HTML 格式标签和实际文本内容。
- 输入文本字段：其内容可通过用户输入也可以通过 ActionScript 进行更改的文本字段。
- 字距调整：调整字符对之间的间距，使单词中的间距比例更佳，文字更易于阅读。
- 静态文本字段：在创作工具中创建的文本字段，运行 SWF 文件时不能更改其内容。
- 文本行量度：文本字段中文本内容不同部分的大小的量度，如文本的基线、字符顶部的高度、下行字符（某些小写字母延伸到基线以下的部分）的大小等等。
- 间距：调整字母组或文本块之间的间距，增大或减小密度，使文字更易于阅读。

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。几乎所有代码清单都涉及到操作文本对象。该对象可能是在 Flash 创作工具中创建并放置在舞台上的，也可能是使用 ActionScript 创建的。测试范例将涉及在 Flash Player 或 AIR 中查看结果，以了解代码对文本或 TextField 对象的影响。

本主题中的示例分为两组。一种类型的示例操作 TextField 对象而不显式创建该对象。若要测试这些代码清单，请执行下面的步骤：

- 1 创建一个空的 Flash 文档。
- 2 在时间轴上选择一个关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 使用“文本”工具在舞台上创建一个动态文本字段。
- 5 使文本字段保持选中，在“属性”检查器中为它指定一个实例名称。该名称应与示例代码清单中文本字段所使用的名称相符，例如，如果代码清单操作名为 myTextField 的文本字段，则应将文本字段也命名为 myTextField。
- 6 通过“控制”>“测试影片”运行程序。

在屏幕上，您将看到按照代码清单所指定的要求操作文本字段的结果。

另一种类型的示例代码清单包含一个旨在用作 SWF 文件的文档类的类定义。在这类代码清单中，文本对象由示例代码创建，因此您不需要另行创建它们。要测试此类型的代码清单，请执行以下操作：

- 1 创建一个空的 Flash 文档并将它保存到您的计算机上。
- 2 创建一个新的 ActionScript 文件，并将它保存到 Flash 文档所在的目录中。文件名应与代码清单中类的名称一致。例如，如果代码清单定义一个名为 TextFieldTest 的类，则使用名称 TextFieldTest.as 来保存 ActionScript 文件。
- 3 将代码清单复制到 ActionScript 文件中并保存该文件。
- 4 在 Flash 文档中，选择“窗口”->“属性”，激活文档“属性”检查器。
- 5 在“属性”检查器的“类”字段中，输入复制自文本的 ActionScript 类的名称。
- 6 通过“控制”>“测试影片”运行程序。

查看屏幕上显示的示例的结果。

用于测试示例代码清单的其它方法在第 29 页的“[测试本章内的示例代码清单](#)”中有更详细的介绍。

使用 **TextField** 类

TextField 类是 Adobe Flex 框架和 Flash 创作环境中提供的其它基于文本的组件（如 **TextArea** 组件或 **TextInput** 组件）的基础。有关在 Flash 创作环境中使用文本组件的详细信息，请参阅《使用 Flash》中的“关于文本控件”。

文本字段内容可以在 SWF 文件中预先指定、从文本文件或数据库中加载，或由用户在与应用程序交互时输入。在文本字段内，文本可以显示为呈现的 HTML 内容，并可在其中嵌入图像。创建文本字段的实例后，可以使用 **flash.text** 类（如 **TextFormat** 和 **StyleSheet**）来控制文本的外观。下面的 **flash.text** 包中涵盖几乎所有与使用 ActionScript 创建文本、管理文本及设置文本格式有关的类。

可以用 **TextFormat** 对象定义格式设置并将此对象分配给文本字段，以此来设置文本格式。如果文本字段包含 HTML 文本，则可以对文本字段应用 **StyleSheet** 对象，以便将样式分配给文本字段内容的特定片段。**TextFormat** 对象或 **StyleSheet** 对象包含定义文本外观（例如颜色、大小和粗细）的属性。**TextFormat** 对象可以将属性分配给文本字段中的所有内容，也可以分配给某个范围的文本。例如，在同一文本字段中，一个句子可以是粗体的红色文本，而下一个句子可以是斜体的蓝色文本。

有关文本格式的详细信息，请参阅第 367 页的“[指定文本格式](#)”。

有关文本字段中 HTML 文本的详细信息，请参阅第 362 页的“[显示 HTML 文本](#)”。

有关样式的详细信息，请参阅第 367 页的“[应用层叠样式表](#)”。

除了 **flash.text** 包中的类以外，您还可以使用 **flash.events.TextEvent** 类响应与文本相关的用户操作。

显示文本

尽管诸如 Adobe Flex Builder 和 Flash 等创作工具为显示文本提供了几种不同的选择（包括与文本相关的组件或文本工具），但以编程方式显示文本的主要途径还是通过文本字段。

文本类型

文本字段中文本的类型根据其来源进行划分：

- 动态文本

动态文本包含从外部源（例如文本文件、XML 文件以及远程 Web 服务）加载的内容。

- 输入文本

输入文本是指用户输入的任何文本或用户可以编辑的动态文本。可以设置样式表来设置输入文本的格式，或使用 **flash.text.TextFormat** 类为输入内容指定文本字段的属性。有关详细信息，请参阅第 365 页的“[捕获文本输入](#)”。

- 静态文本

静态文本只能通过创作工具来创建。不能使用 ActionScript 3.0 创建静态文本实例。但是，可以使用 ActionScript 类（例如 **StaticText** 和 **TextSnapshot**）来操作现有的静态文本实例。有关详细信息，请参阅第 371 页的“[处理静态文本](#)”。

修改文本字段内容

可通过将字符串指派给 **flash.text.TextField.text** 属性来定义动态文本。可以直接将字符串赋予该属性，如下所示：

```
myTextField.text = "Hello World";
```

还可以为 **text** 属性赋予在脚本中定义的变量值，如下例所示：

```
package
{
    import flash.display.Sprite;
    import flash.text.*;

    public class TextWithImage extends Sprite
    {
        private var myTextBox:TextField = new TextField();
        private var myText:String = "Hello World";

        public function TextWithImage()
        {
            addChild(myTextBox);
            myTextBox.text = myText;
        }
    }
}
```

或者，可以将一个远程变量的值赋予 `text` 属性。从远程源加载文本值有三种方式：

- `flash.net.URLLoader` 和 `flash.net.URLRequest` 类可以从本地或远程位置为文本加载变量。
- `FlashVars` 属性被嵌入到承载 SWF 文件的 HTML 页中，可以包含文本变量的值。
- `flash.net.SharedObject` 类管理值的永久存储。有关详细信息，请参阅第 522 页的“[存储本地数据](#)”。

显示 HTML 文本

`flash.text.TextField` 类具有一个 `htmlText` 属性，可使用它将您的文本字符串标识为包含用于设置内容格式的 HTML 标签。如下例所示，必须将您的字符串值赋予 `htmlText` 属性（而不是 `text` 属性），以便 Flash Player 或 AIR 将文本呈现为 HTML：

```
var myText:String = "<p>This is <b>some</b> content to <i>render</i> as <u>HTML</u> text.</p>";
myTextBox.htmlText = myText;
```

Flash Player 和 AIR 支持用于 `htmlText` 属性的 HTML 标签和实体的一个子集。《ActionScript 3.0 语言和组件参考》中的 `flash.text.TextField.htmlText` 属性说明提供了有关受支持的 HTML 标签和实体的详细信息。

一旦您使用 `htmlText` 属性指定了内容，就可以使用样式表或 `textformat` 标签来管理内容的格式设置。有关详细信息，请参阅第 367 页的“[设置文本格式](#)”。

在文本字段中使用图像

将内容显示为 HTML 文本的另一个好处是可以在文本字段中包括图像。可以使用 `img` 标签引用一个本地或远程图像，并使其显示在关联的文本字段内。

以下示例将创建一个名为 `myTextBox` 的文本字段，并在显示的文本中包括一个内容为眼睛的 JPG 图像，该图像与 SWF 文件存储在同一目录下：

```
package
{
    import flash.display.Sprite;
    import flash.text.*;

    public class TextWithImage extends Sprite
    {
        private var myTextBox:TextField;
        private var myText:String = "<p>This is <b>some</b> content to <i>test</i> and <i>see</i></p><p><img src='eye.jpg' width='20' height='20'></p><p>what can be rendered.</p><p>You should see an eye image and some <u>HTML</u> text.</p>";
        public function TextWithImage()
        {
            myTextBox.width = 200;
            myTextBox.height = 200;
            myTextBox.multiline = true;
            myTextBox.wordWrap = true;
            myTextBox.border = true;

            addChild(myTextBox);
            myTextBox.htmlText = myText;
        }
    }
}
```

img 标签支持 JPEG、GIF、PNG 和 SWF 文件。

在文本字段中滚动文本

在许多情况下，文本可能比显示该文本的文本字段长。或者，某个输入字段允许用户输入比字段一次可显示的文本内容更多的文本。您可以使用 `flash.text.TextField` 类的与滚动相关的属性来管理过长的内容（垂直或水平方向）。

与滚动有关的属性包括 `TextField.scrollV`、`TextField.scrollH`、`maxScrollV` 和 `maxScrollH`。可使用这些属性来响应鼠标单击或按键等事件。

以下示例将创建一个已设置大小的文本字段，其中包含的文本内容超过了该字段一次可以显示的内容。用户单击文本字段时，文本会在垂直方向上滚动。

```
package
{
    import flash.display.Sprite;
    import flash.text.*;
    import flash.events.MouseEvent;

    public class TextScrollExample extends Sprite
    {
        private var myTextBox:TextField = new TextField();
        private var myText:String = "Hello world and welcome to the show. It's really nice to meet you.
Take your coat off and stay a while. OK, show is over. Hope you had fun. You can go home now. Don't forget
to tip your waiter. There are mints in the bowl by the door. Thank you. Please come again.";

        public function TextScrollExample()
        {
            myTextBox.text = myText;
            myTextBox.width = 200;
            myTextBox.height = 50;
            myTextBox.multiline = true;
            myTextBox.wordWrap = true;
            myTextBox.background = true;
            myTextBox.border = true;

            var format:TextFormat = new TextFormat();
            format.font = "Verdana";
            format.color = 0xFF0000;
            format.size = 10;

            myTextBox.defaultTextFormat = format;
            addChild(myTextBox);
            myTextBox.addEventListener(MouseEvent.MOUSE_DOWN, mouseDownScroll);
        }

        public function mouseDownScroll(event:MouseEvent):void
        {
            myTextBox.scrollV++;
        }
    }
}
```

选择和操作文本

您可以选择动态文本或输入文本。由于 `TextField` 类的文本选择属性和方法使用索引位置来设置要操作的文本的范围，因此即使不知道内容，您也可以以编程方式选择动态文本或输入文本。

注: 在 Flash 创作工具中，如果对静态文本字段选择了可选选项，则导出并置于显示列表中的文本字段为常规的动态文本字段。

选择文本

默认情况下，`flash.text.TextField.selectable` 属性为 `true`，您可以使用 `setSelection()` 方法以编程方式选择文本。

例如，您可以将某个文本字段中的特定文本设置成用户单击该文本字段时处于选定状态：

```
var myTextField:TextField = new TextField();
myTextField.text = "No matter where you click on this text field the TEXT IN ALL CAPS is selected.";
myTextField.autoSize = TextFieldAutoSize.LEFT;
addChild(myTextField);
addEventListener(MouseEvent.CLICK, selectText);

function selectText(event:MouseEvent):void
{
    myTextField.setSelection(49, 65);
}
```

同样，如果您想让文本字段中的文本一开始显示时就处于选定状态，可以创建一个在向显示列表中添加该文本字段时调用的事件处理函数。

捕获用户选择的文本

TextField 的 selectionBeginIndex 和 selectionEndIndex 属性可用于捕获用户当前选择的任何内容，这两个属性为“只读”属性，因此不能设置为以编程方式选择文本。此外，输入文本字段也可以使用 caretIndex 属性。

例如，以下代码将跟踪用户所选文本的索引值：

```
var myTextField:TextField = new TextField();
myTextField.text = "Please select the TEXT IN ALL CAPS to see the index values for the first and last
letters.";
myTextField.autoSize = TextFieldAutoSize.LEFT;
addChild(myTextField);
addEventListener(MouseEvent.MOUSE_UP, selectText);

function selectText(event:MouseEvent):void
{
    trace("First letter index position: " + myTextField.selectionBeginIndex);
    trace("Last letter index position: " + myTextField.selectionEndIndex);
}
```

您可以对所选内容应用 TextFormat 对象属性的集合来更改文本的外观。有关将 TextFormat 属性的集合应用于选定文本的详细信息，请参阅第 369 页的“[设置文本字段内文本范围的格式](#)”。

捕获文本输入

默认情况下，文本字段的 type 属性设置为 dynamic。如果使用 TextFieldType 类将 type 属性设置为 input，则可以收集用户输入并保存该值以便在应用程序的其它部分使用。对于表单以及希望用户定义可用于程序中其它位置的文本值的任何应用程序而言，输入文本字段都十分有用。

例如，以下代码会创建一个名为 myTextBox 的输入文本字段。当用户在字段中输入文本时，会触发 textInput 事件。名为 textInputCapture 的事件处理函数会捕获输入的文本字符串，并将其赋予一个变量。Flash Player 或 AIR 会在另一个名为 myOutputBox 的文本字段中显示新文本。

```

package
{
    import flash.display.Sprite;
    import flash.display.Stage;
    import flash.text.*;
    import flash.events.*;

    public class CaptureUserInput extends Sprite
    {
        private var myTextBox:TextField = new TextField();
        private var myOutputBox:TextField = new TextField();
        private var myText:String = "Type your text here.";

        public function CaptureUserInput()
        {
            captureText();
        }

        public function captureText():void
        {
            myTextBox.type = TextFieldType.INPUT;
            myTextBox.background = true;
            addChild(myTextBox);
            myTextBox.text = myText;
            myTextBox.addEventListener(TextEvent.TEXT_INPUT, textInputCapture);
        }

        public function textInputCapture(event:TextEvent):void
        {
            var str:String = myTextBox.text;
            createOutputBox(str);
        }

        public function createOutputBox(str:String):void
        {
            myOutputBox.background = true;
            myOutputBox.x = 200;
            addChild(myOutputBox);
            myOutputBox.text = str;
        }
    }
}

```

限制文本输入

由于输入文本字段经常用于表单或应用程序中的对话框，所以您可能想要限制用户在文本字段中输入的字符的类型，或者甚至希望将文本隐藏（例如，密码文本）。可以设置 `flash.text.TextField` 类的 `displayAsPassword` 属性和 `restrict` 属性来控制用户输入。

`displayAsPassword` 属性只是在用户键入文本时将其隐藏（显示为一系列星号）。当 `displayAsPassword` 设置为 `true` 时，“剪切”和“复制”命令及其对应的键盘快捷键将不起作用。如下例所示，为 `displayAsPassword` 属性赋值的过程与为其它属性（如背景和颜色）赋值类似：

```

myTextBox.type = TextFieldType.INPUT;
myTextBox.background = true;
myTextBox.displayAsPassword = true;
addChild(myTextBox);

```

`restrict` 属性则更为复杂一些，因为您需要指定允许用户在输入文本字段中键入哪些字符。可以允许特定字母、数字或字母、数字和字符的范围。以下代码只允许用户在文本字段中输入大写字母（不包括数字或特殊字符）：

```
myTextBox.restrict = "A-Z";
```

ActionScript 3.0 使用连字符来定义范围，使用尖号来定义被排除的字符。有关定义输入文本字段中受限制的内容的详细信息，请参阅《ActionScript 3.0 语言和组件参考》中的 `flash.text.TextField.restrict` 属性条目。

设置文本格式

以编程方式设置文本显示的格式设置有多种方式。可以直接在 `TextField` 实例中设置属性，例如，`TextField.thickness`、`TextField.textColor` 和 `TextField.textHeight` 属性。也可以使用 `htmlText` 属性指定文本字段的内容，并使用受支持的 HTML 标签，如 `b`、`i` 和 `u`。但是您也可以将 `TextFormat` 对象应用于包含纯文本的文本字段，或将 `StyleSheet` 对象应用于包含 `htmlText` 属性的文本字段。使用 `TextFormat` 和 `StyleSheet` 对象可以对整个应用程序的文本外观提供最有力的控制和最佳的一致性。可以定义 `TextFormat` 或 `StyleSheet` 对象并将其应用于应用程序中的部分或所有文本字段。

指定文本格式

您可以使用 `TextFormat` 类设置多个不同的文本显示属性，并将它们应用于 `TextField` 对象的整个内容或一定范围的文本。

以下示例对整个 `TextField` 对象应用一个 `TextFormat` 对象，并对 `TextField` 对象中一定范围的文本应用另一个 `TextFormat` 对象：

```
var tf:TextField = new TextField();
tf.text = "Hello Hello";

var format1:TextFormat = new TextFormat();
format1.color = 0xFF0000;

var format2:TextFormat = new TextFormat();
format2.font = "Courier";

tf.setTextFormat(format1);
var startRange:uint = 6;
tf.setTextFormat(format2, startRange);

addChild(tf);
```

`TextField.setTextFormat()` 方法只影响已显示在文本字段中的文本。如果 `TextField` 中的内容发生更改，则应用程序可能需要重新调用 `TextField.setTextFormat()` 方法以便重新应用格式设置。您也可以设置 `TextField` 的 `defaultTextFormat` 属性来指定用户输入文本所用的格式。

应用层叠样式表

文本字段可以包含纯文本或 HTML 格式的文本。纯文本存储在实例的 `text` 属性中，而 HTML 文本存储在 `htmlText` 属性中。

您可以使用 CSS 样式声明来定义可应用于多种不同文本字段的文本样式。CSS 样式声明可以在应用程序代码中进行创建，也可以在运行时从外部 CSS 文件中加载。

`flash.text.StyleSheet` 类用于处理 CSS 样式。`StyleSheet` 类可识别有限的 CSS 属性集合。有关 `StyleSheet` 类支持的样式属性的详细列表，请参阅《ActionScript 3.0 语言和组件参考》中的 `flash.textStylesheet` 条目。

如以下示例所示，您可以在代码中创建 CSS，并使用 `StyleSheet` 对象对 HTML 文本应用这些样式：

```

var style:StyleSheet = new StyleSheet();

var styleObj:Object = new Object();
styleObj.fontSize = "bold";
styleObj.color = "#FF0000";
style.setStyle(".darkRed", styleObj);

var tf:TextField = new TextField();
tf.styleSheet = style;
tf.htmlText = "<span class = 'darkRed'>Red</span> apple";

addChild(tf);

```

创建 `StyleSheet` 对象后，示例代码创建一个简单对象以容纳一组样式声明属性。然后该代码调用 `StyleSheet.setStyle()` 方法，该方法将名为 `".darkred"` 的新样式添加到样式表中。接着，代码通过将 `StyleSheet` 对象分配给 `TextField` `styleSheet` 属性来应用样式表格式设置。

要使 CSS 样式生效，应在设置 `htmlText` 属性之前对 `TextField` 对象应用样式表。

根据设计，带有样式表的文本字段是不可编辑的。如果您有一个输入文本字段并为其分配一个样式表，则该文本字段将显示样式表的属性，但不允许用户在其中输入新的文本。而且，您也无法在分配有样式表的文本字段上使用以下 ActionScript API：

- `TextField.replaceText()` 方法
- `TextField.replaceSelectedText()` 方法
- `TextField.defaultTextFormat` 属性
- `TextField.setTextFormat()` 方法

如果某个文本字段已经分配了一个样式表，但后来将 `TextField.styleSheet` 属性设置为 `null`，则 `TextField.text` 和 `TextField.htmlText` 属性的内容会向它们的内容中添加标签和属性，以结合先前分配的样式表设定的格式。若要保留原始 `htmlText` 属性，应在将样式表设置为 `null` 之前将其保存在变量中。

加载外部 CSS 文件

用于设置格式的 CSS 方法的功能更加强大，您可以在运行时从外部文件加载 CSS 信息。当 CSS 数据位于应用程序本身以外时，您可以更改应用程序中的文本的可视样式，而不必更改 ActionScript 3.0 源代码。部署完应用程序后，可以通过更改外部 CSS 文件来更改应用程序的外观，而不必重新部署应用程序 SWF 文件。

`StyleSheet.parseCSS()` 方法可将包含 CSS 数据的字符串转换为 `StyleSheet` 对象中的样式声明。以下示例显示如何读取外部 CSS 文件并对 `TextField` 对象应用其样式声明。

首先，下面是要加载的 CSS 文件（名为 `example.css`）的内容：

```

p {
    font-family: Times New Roman, Times, _serif;
    font-size: 14;
}

h1 {
    font-family: Arial, Helvetica, _sans;
    font-size: 20;
    font-weight: bold;
}

.bluetext {
    color: #0000CC;
}

```

接下来是加载该 `example.css` 文件并对 `TextField` 内容应用样式的类的 ActionScript 代码：

```
package
{
    import flash.display.Sprite;
    import flash.events.Event;
    import flash.net.URLLoader;
    import flash.net.URLRequest;
    import flash.text.StyleSheet;
    import flash.text.TextField;
    import flash.text.TextFieldAutoSize;

    public class CSSFormattingExample extends Sprite
    {
        var loader:URLLoader;
        var field:TextField;
        var exampleText:String = "<h1>This is a headline</h1>" +
            "<p>This is a line of text. <span class='bluetext'>" +
            "This line of text is colored blue.</span></p>";

        public function CSSFormattingExample():void
        {
            field = new TextField();
            field.width = 300;
            field.autoSize = TextFieldAutoSize.LEFT;
            field.wordWrap = true;
            addChild(field);

            var req:URLRequest = new URLRequest("example.css");

            loader = new URLLoader();
            loader.addEventListener(Event.COMPLETE, onCSSFileLoaded);
            loader.load(req);
        }

        public function onCSSFileLoaded(event:Event):void
        {
            var sheet:StyleSheet = new StyleSheet();
            sheet.parseCSS(loader.data);
            field.styleSheet = sheet;
            field.htmlText = exampleText;
        }
    }
}
```

加载 CSS 数据后，会执行 `onCSSFileLoaded()` 方法并调用 `StyleSheet.parseCSS()` 方法，将样式声明传送给 `StyleSheet` 对象。

设置文本字段内文本范围的格式

`flash.text.TextField` 类的一个特别有用的方法是 `setTextFormat()` 方法。使用 `setTextFormat()`，您可以将特定属性分配给文本字段的部分内容以响应用户输入，例如，需要提醒用户必须输入特定条目的表单，或在用户选择部分文本时提醒用户更改文本字段内文本段落小节的重点的表单。

以下示例对某一范围的字符使用 `TextField.setTextFormat()`，以便在用户单击文本字段时更改 `myTextField` 的部分内容的外观：

```
var myTextField:TextField = new TextField();
myTextField.text = "No matter where you click on this text field the TEXT IN ALL CAPS changes format.";
myTextField.autoSize = TextFieldAutoSize.LEFT;
addChild(myTextField);
addEventListener(MouseEvent.CLICK, changeText);

var myformat:TextFormat = new TextFormat();
myformat.color = 0xFF0000;
myformat.size = 18;
myformat.underline = true;

function changeText(event:MouseEvent):void
{
    myTextField.setTextFormat(myformat, 49, 65);
}
```

高级文本呈现

ActionScript 3.0 在 `flash.text` 包中提供多个类来控制所显示文本的属性，包括嵌入字体、消除锯齿设置、`alpha` 通道控制及其它特定设置。《ActionScript 3.0 语言和组件参考》提供了有关这些类和属性（包括 `CSMSettings`、`Font` 和 `TextRenderer` 类）的详细说明。

使用嵌入字体

当您在应用程序中为 `TextField` 指定特定字体时，Flash Player 或 AIR 会查找同名的设备字体（位于用户计算机上的一种字体）。如果在系统上没有找到该字体，或者如果用户的字体版本与具有该名称的字体略有差异，则文本显示外观会与预想的情况差别很大。

若要确保用户看到完全正确的字体，您可以将该字体嵌入到应用程序 SWF 文件中。嵌入字体有很多好处：

- 嵌入字体字符是消除锯齿的，特别是对于较大的文本，该字体可以使文本边缘看起来更平滑。
- 可以旋转使用嵌入字体的文本。
- 嵌入字体文本可以产生透明或半透明效果。
- 可以对嵌入字体使用字距调整的 CSS 样式。

使用嵌入字体的最大限制是嵌入字体会增加文件大小或应用程序的下载大小。

将字体文件嵌入到应用程序 SWF 文件中的具体方法因开发环境而异。

嵌入字体后，可以确保 `TextField` 使用正确的嵌入字体：

- 将 `TextField` 的 `embedFonts` 属性设置为 `true`。
- 创建一个 `TextFormat` 对象，将其 `fontFamily` 属性设置为嵌入字体的名称，并对 `TextField` 应用 `TextFormat` 对象。指定嵌入字体时，`fontFamily` 属性应只包含一个名称；该名称不能是用逗号分隔的由多个字体名称构成的列表。
- 如果使用 CSS 样式为 `TextField` 或组件设置字体，请将 `font-family` CSS 属性设置为嵌入字体的名称。如果要指定一种嵌入字体，则 `font-family` 属性必须包含单一名称，而不能是多个名称的列表。

在 Flash 中嵌入字体

Flash 创作工具可嵌入系统中已安装的几乎所有字体，包括 TrueType 字体和 Type 1 Postscript 字体。

可以利用多种方法将字体嵌入应用程序，其中包括：

- 在舞台上设置 `TextField` 的字体和样式属性，然后单击“嵌入字体”复选框
- 创建并引用字体元件
- 创建并使用包含嵌入字体元件的运行时共享库

有关如何在应用程序中嵌入字体的详细信息，请参阅《使用 Flash》中的“动态或输入文本字段的嵌入字体”。

控制清晰度、粗细和消除锯齿

默认情况下，在文本调整大小、更改颜色或在不同背景上显示时，Flash Player 或 AIR 可以确定文本显示控件的设置（如清晰度、粗细和消除锯齿）。在某些情况下，如文本很小、很大或显示在各种特别的背景上时，您可能需要保持对这些设置的控制。可以使用 `flash.text.TextRenderer` 类及其相关类（如 `CSMSettings` 类）来覆盖 Flash Player 或 AIR 的设置。使用这些类可以精确控制嵌入文本的呈现品质。有关嵌入字体的详细信息，请参阅第 370 页的“[使用嵌入字体](#)”。

注：为了设置清晰度、粗细或 `gridFitType` 属性，或者使用 `TextRenderer.setAdvancedAntiAliasingTable()` 方法，`flash.text.TextField.antiAliasType` 属性的值必须是 `AntiAliasType.ADVANCED`。

以下示例使用名为 `myFont` 的嵌入字体对显示的文本应用自定义连续笔触调制（CSM）属性和格式设置。用户单击显示的文本时，Flash Player 或 Adobe AIR 会应用自定义设置：

```
var format:TextFormat = new TextFormat();
format.color = 0x336699;
format.size = 48;
format.font = "myFont";

var myText:TextField = new TextField();
myText.embedFonts = true;
myText.autoSize = TextFieldAutoSize.LEFT;
myText.antiAliasType = AntiAliasType.ADVANCED;
myText.defaultTextFormat = format;
myText.selectable = false;
myText.mouseEnabled = true;
myText.text = "Hello World";
addChild(myText);
myText.addEventListener(MouseEvent.CLICK, clickHandler);

function clickHandler(event:Event):void
{
    var myAntiAliasSettings = new CSMSettings(48, 0.8, -0.8);
    var myAliasTable:Array = new Array(myAntiAliasSettings);
    TextRenderer.setAdvancedAntiAliasingTable("myFont", FontStyle.ITALIC, TextColorType.DARK_COLOR,
myAliasTable);
}
```

处理静态文本

静态文本只能在创作工具中创建。不能使用 ActionScript 以编程方式对静态文本进行实例化。静态文本非常适用于简短且不会更改（动态文本则会更改）的文本。可以将静态文本看作一种图形元素，类似于通过 Flash 创作工具在舞台上绘制的圆形或正方形。由于静态文本比动态文本受到更多的限制，ActionScript 3.0 允许使用 `StaticText` 类读取静态文本的属性值。另外还可使用 `TextSnapshot` 类从静态文本中读取值。

使用 `StaticText` 类访问静态文本字段

通常，可以在 Flash 创作工具的“动作”面板中使用 `flash.text.StaticText` 类来与放置在舞台上的静态文本实例进行交互。也可以在与包含静态文本的 SWF 文件进行交互的 ActionScript 文件中执行类似工作。但是这两种情况下都不能以编程方式对静态文本实例进行实例化。静态文本是在创作工具中创建的。

若要创建对现有静态文本字段的引用，可以遍历显示列表中的项目并分配一个变量。例如：

```

for (var i = 0; i < this.numChildren; i++) {
    var displayitem:DisplayObject = this.getChildAt(i);
    if (displayitem instanceof StaticText) {
        trace("a static text field is item " + i + " on the display list");
        var myFieldLabel:StaticText = StaticText(displayitem);
        trace("and contains the text: " + myFieldLabel.text);
    }
}

```

引用某个静态文本字段之后，您可以在 ActionScript 3.0 中使用该字段的属性。下面的代码附加到时间轴上的一个帧，并假设一个静态文本引用分配有一个名为 myFieldLabel 的变量。相对于 myFieldLabel 的 x 和 y 值放置名为 myField 的动态文本字段，并再次显示 myFieldLabel 的值。

```

var myField:TextField = new TextField();
addChild(myField);
myField.x = myFieldLabel.x;
myField.y = myFieldLabel.y + 20;
myField.autoSize = TextFieldAutoSize.LEFT;
myField.text = "and " + myFieldLabel.text

```

使用 TextSnapshot 类

如果要以编程方式使用现有静态文本实例，可以使用 `flash.text.TextSnapshot` 类来与 `flash.display.DisplayObjectContainer` 的 `textSnapshot` 属性配合工作。也就是说，通过 `DisplayObjectContainer.textSnapshot` 属性创建 `TextSnapshot` 实例。然后，可以将方法应用于该实例，以检索值或选择部分静态文本。

例如，请在舞台上放置一个包含文本“TextSnapshot Example”的静态文本字段。将下面的 ActionScript 添加到时间轴中的第 1 帧：

```

var mySnap:TextSnapshot = this.textSnapshot;
var count:Number = mySnap.charCount;
mySnap.setSelected(0, 4, true);
mySnap.setSelected(1, 2, false);
var myText:String = mySnap.getSelectedText(false);
trace(myText);

```

如果要在应用程序的其它部分将该文本作为值使用，则 `TextSnapshot` 类对于从所加载的 SWF 文件中的静态文本字段中获取文本非常有用。

TextField 示例：报纸风格的文本格式设置

“新闻布局”示例设置文本的格式，使文本的外观看起来有点象印刷报纸中的素材。输入文本可以包含标题、副标题和素材正文。在给定显示宽度和高度的情况下，此“新闻布局”示例将会设置标题和副标题的格式，使其占据整个显示区域的宽度。素材文本分布在两列或多列中。

此示例演示以下 ActionScript 编程技巧：

- 扩展 `TextField` 类
- 加载并应用外部 CSS 文件
- 将 CSS 样式转换为 `TextFormat` 对象
- 使用 `TextLineMetrics` 类获取有关文本显示大小的信息

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。“新闻布局”应用程序文件位于 `Samples/NewsLayout` 文件夹中。该应用程序包含以下文件：

文件	描述
NewsLayout.mxml 或 NewsLayout.fla	适用于 Flex (MXML) 或 Flash (FLA) 的应用程序的用户界面。
com/example/programmingas3/ne wslayout/StoryLayoutComponent.a s	放置 StoryLayout 实例的 Flex UIComponent 类。
com/example/programmingas3/ne wslayout/StoryLayout.as	排列用于显示的所有新闻素材组件的主要 ActionScript 类。
com/example/programmingas3/ne wslayout/FormattedTextField.as	管理本身的 TextFormat 对象的 TextField 类的子类。
com/example/programmingas3/ne wslayout/HeadlineTextField.as	调整字体大小以适合需要的宽度的 FormattedTextField 类的子类。
com/example/programmingas3/ne wslayout/MultiColumnTextField.as	在两列或多列之间拆分文本的 ActionScript 类。
story.css	为布局定义文本样式的 CSS 文件。

读取外部 CSS 文件

“新闻布局”应用程序开始时读取本地 XML 文件中的素材文本。然后，它读取提供标题、副标题和主体文本的格式设置信息的外部 CSS 文件。

CSS 文件定义三种样式：用于素材的标准段落样式和分别用于标题和副标题的 h1 和 h2 样式。

```

p {
    font-family: Georgia, "Times New Roman", Times, _serif;
    font-size: 12;
    leading: 2;
    text-align: justify;
    indent: 24;
}

h1 {
    font-family: Verdana, Arial, Helvetica, _sans;
    font-size: 20;
    font-weight: bold;
    color: #000099;
    text-align: left;
}

h2 {
    font-family: Verdana, Arial, Helvetica, _sans;
    font-size: 16;
    font-weight: normal;
    text-align: left;
}

```

用于读取外部 CSS 文件的方法与第 368 页的“[加载外部 CSS 文件](#)”中所述的方法相同。加载 CSS 文件后，应用程序执行 onCSSFileLoaded() 方法，如下所示。

```
public function onCSSFileLoaded(event:Event):void
{
    this.sheet = new StyleSheet();
    this.sheet.parseCSS(loader.data);

    h1Format = getTextStyle("h1", this.sheet);
    if (h1Format == null)
    {
        h1Format = getDefaultHeadFormat();
    }
    h2Format = getTextStyle("h2", this.sheet);
    if (h2Format == null)
    {
        h2Format = getDefaultHeadFormat();
        h2Format.size = 16;
    }
    pFormat = getTextStyle("p", this.sheet);
    if (pFormat == null)
    {
        pFormat = getDefaultTextFormat();
        pFormat.size = 12;
    }
    displayText();
}
```

onCSSFileLoaded() 方法创建一个 StyleSheet 对象并使之分析输入 CSS 数据。素材的主体文本将显示在 MultiColumnTextField 对象中，该对象可以直接使用 StyleSheet 对象。不过，标题字段使用 HeadlineTextField 类，该类使用 TextFormat 对象进行格式设置。

onCSSFileLoaded() 方法调用两次 getTextStyle() 方法，将 CSS 样式声明转换为 TextFormat 对象，以便与两个 HeadlineTextField 对象中的每个对象配合使用。

```
public function getTextStyle(styleName:String, ss:StyleSheet):TextFormat
{
    var format:TextFormat = null;

    var style:Object = ss.getStyle(styleName);
    if (style != null)
    {
        var colorStr:String = style.color;
        if (colorStr != null && colorStr.indexOf("#") == 0)
        {
            style.color = colorStr.substr(1);
        }
        format = new TextFormat(style.fontFamily,
                               style.fontSize,
                               style.color,
                               (style.fontWeight == "bold"),
                               (style.fontStyle == "italic"),
                               (style.textDecoration == "underline"),
                               style.url,
                               style.target,
                               style.textAlign,
                               style.marginLeft,
                               style.marginRight,
                               style.indent,
                               style.leading);

        if (style.hasOwnProperty("letterSpacing"))
        {
            format.letterSpacing = style.letterSpacing;
        }
    }
    return format;
}
```

CSS 样式声明和 **TextFormat** 对象的属性名称和属性值的含义不同。**getTextStyle()** 方法可以将 CSS 属性值转换为 **TextFormat** 对象预期的值。

在页面上排列素材元素

StoryLayout 类可以设置标题、副标题和主体文本字段的格式并将这些字段的布局排列为报纸样式。**displayText()** 方法一开始会创建并放置各个字段。

```
public function displayText():void
{
    headlineTxt = new HeadlineTextField(h1Format);
    headlineTxt.wordWrap = true;
    headlineTxt.x = this.paddingLeft;
    headlineTxt.y = this.paddingTop;
    headlineTxt.width = this.preferredWidth;
    this.addChild(headlineTxt);

    headlineTxt.fitText(this.headline, 1, true);

    subtitleTxt = new HeadlineTextField(h2Format);
    subtitleTxt.wordWrap = true;
    subtitleTxt.x = this.paddingLeft;
    subtitleTxt.y = headlineTxt.y + headlineTxt.height;
    subtitleTxt.width = this.preferredWidth;
    this.addChild(subtitleTxt);

    subtitleTxt.fitText(this.subtitle, 2, false);

    storyTxt = new MultiColumnText(this.numColumns, 20,
        this.preferredWidth, 400, true, this.pFormat);
    storyTxt.x = this.paddingLeft;
    storyTxt.y = subtitleTxt.y + subtitleTxt.height + 10;
    this.addChild(storyTxt);

    storyTxt.text = this.content;
    ...
}
```

将一个字段的 `y` 属性设置为等于上一个字段的 `y` 属性加上其自身高度，这样即可将每个字段放置在上一个字段的下面。由于 `HeadlineTextField` 对象和 `MultiColumnText` 对象可以更改其高度以适应内容，因此需要进行这种动态位置计算。

更改字体大小以适合字段大小

在给定要显示内容的宽度（以像素为单位）和最多行数的情况下，`HeadlineTextField` 会更改字体大小以使文本适合字段大小。如果文本很短，则字体大小很大，从而产生小报样式的标题。如果文本很长，那么字体大小较小。

下面所示的 `HeadlineTextField.fitText()` 方法的作用就是调整字体大小：

```
public function fitText(msg:String, maxLines:uint = 1, toUpper:Boolean = false, targetWidth:Number = -1):uint
{
    this.text = toUpper ? msg.toUpperCase() : msg;

    if (targetWidth == -1)
    {
        targetWidth = this.width;
    }

    var pixelsPerChar:Number = targetWidth / msg.length;

    var pointSize:Number = Math.min(MAX_POINT_SIZE, Math.round(pixelsPerChar * 1.8 * maxLines));

    if (pointSize < 6)
    {
        // the point size is too small
        return pointSize;
    }

    this.changeSize(pointSize);

    if (this.numLines > maxLines)
```

```

    {
        return shrinkText(--pointSize, maxLines);
    }
    else
    {
        return growText(pointSize, maxLines);
    }
}

public function growText(pointSize:Number, maxLines:uint = 1):Number
{
    if (pointSize >= MAX_POINT_SIZE)
    {
        return pointSize;
    }

    this.changeSize(pointSize + 1);

    if (this.numLines > maxLines)
    {
        // set it back to the last size
        this.changeSize(pointSize);
        return pointSize;
    }
    else
    {
        return growText(pointSize + 1, maxLines);
    }
}

public function shrinkText(pointSize:Number, maxLines:uint=1):Number
{
    if (pointSize <= MIN_POINT_SIZE)
    {
        return pointSize;
    }

    this.changeSize(pointSize);

    if (this.numLines > maxLines)
    {
        return shrinkText(pointSize - 1, maxLines);
    }
    else
    {
        return pointSize;
    }
}

```

`HeadlineTextField.fitText()` 方法使用简单的递归技术来调整字体大小。首先，该方法推测文本中每个字符的平均像素数，并据此计算起始磅值。然后，它更改字体大小并检查文本中的文字是否换行，从而产生比最大值更多的文本行。如果文本行过多，则它会调用 `shrinkText()` 方法减小字体大小并重试。如果文本行不太多，则它会调用 `growText()` 方法增大字体大小并重试。当字体大小再增加一磅即会产生过多行时，该过程即会停止。

在多列之间拆分文本

`MultiColumnTextField` 类会在多个 `TextField` 对象中分布文本，然后将这些对象排列成报纸专栏的样式。

`MultiColumnTextField()` 构造函数首先创建一个由 `TextField` 对象构成的数组，每列一个对象，如下所示：

```

for (var i:int = 0; i < cols; i++)
{
    var field:TextField = new TextField();
    field.multiline = true;
    field.autoSize = TextFieldAutoSize.NONE;
    field.wordWrap = true;
    field.width = this.colWidth;
    field.setTextFormat(this.format);
    this.fieldArray.push(field);
    this.addChild(field);
}

```

每个 `TextField` 对象均使用 `addChild()` 方法添加到该数组中，并添加到显示列表中。

只要 `StoryLayout` 的 `text` 属性或 `styleSheet` 属性发生更改，该对象就会调用 `layoutColumns()` 方法来重新显示文本。`layoutColumns()` 方法会调用 `getOptimalHeight()` 方法，以计算在给定布局宽度内适合所有文本所需的高度。

```

public function getOptimalHeight(str:String):int
{
    if (field.text == "" || field.text == null)
    {
        return this.preferredHeight;
    }
    else
    {
        this.linesPerCol = Math.ceil(field.numLines / this.numColumns);

        var metrics:TextLineMetrics = field.getLineMetrics(0);
        this.lineHeight = metrics.height;
        var prefHeight:int = linesPerCol * this.lineHeight;

        return prefHeight + 4;
    }
}

```

首先，`getOptimalHeight()` 方法计算每列的宽度。然后设置数组中第一个 `TextField` 对象的宽度和 `htmlText` 属性。`getOptimalHeight()` 方法使用第一个 `TextField` 对象计算文本中自动换行文本的总行数，并据此确定每列中应有多少行。接下来，它调用 `TextField.getLineMetrics()` 方法以检索 `TextLineMetrics` 对象，该对象包含有关第一行的文本大小的详细信息。`TextLineMetrics.height` 属性用像素表示文本行的完整高度，包括上升、下降和前导。`MultiColumnTextField` 对象的最佳高度即为行高度乘以每列行数再加上 4（`TextField` 对象顶部边框和底部边框各 2 个像素）。

以下是完整 `layoutColumns()` 方法的代码：

```

public function layoutColumns():void
{
    if (this._text == "" || this._text == null)
    {
        return;
    }

    var field:TextField = fieldArray[0] as TextField;
    field.text = this._text;
    field.setTextFormat(this.format);

    this.preferredHeight = this.getOptimalHeight(field);

    var remainder:String = this._text;
    var fieldText:String = "";
    var lastLineEndedPara:Boolean = true;

    var indent:Number = this.format.indent as Number;

    for (var i:int = 0; i < fieldArray.length; i++)

```

```
{  
    field = this.fieldArray[i] as TextField;  
  
    field.height = this.preferredHeight;  
    field.text = remainder;  
  
    field.setTextFormat(this.format);  
  
    var lineLen:int;  
    if (indent > 0 && !lastLineEndedPara && field.numLines > 0)  
    {  
        lineLen = field.getLineLength(0);  
        if (lineLen > 0)  
        {  
            field.setTextFormat(this.firstLineFormat, 0, lineLen);  
        }  
    }  
  
    field.x = i * (colWidth + gutter);  
    field.y = 0;  
  
    remainder = "";  
    fieldText = "";  
  
    var linesRemaining:int = field.numLines;  
    var linesVisible:int = Math.min(this.linesPerCol, linesRemaining);  
  
    for (var j:int = 0; j < linesRemaining; j++)  
    {  
        if (j < linesVisible)  
        {  
            fieldText += field.getLineText(j);  
        }  
        else  
        {  
            remainder += field.getLineText(j);  
        }  
    }  
  
    field.text = fieldText;  
  
    field.setTextFormat(this.format);  
  
    if (indent > 0 && !lastLineEndedPara)  
    {  
        lineLen = field.getLineLength(0);  
        if (lineLen > 0)  
        {  
            field.setTextFormat(this.firstLineFormat, 0, lineLen);  
        }  
    }  
  
    var lastLine:String = field.getLineText(field.numLines - 1);  
    var lastCharCode:Number = lastLine.charCodeAt(lastLine.length - 1);
```

```
if (lastCharCode == 10 || lastCharCode == 13)
{
    lastLineEndedPara = true;
}
else
{
    lastLineEndedPara = false;
}

if ((this.format.align == TextFormatAlign.JUSTIFY) &&
    (i < fieldArray.length - 1))
{
    if (!lastLineEndedPara)
    {
        justifyLastLine(field, lastLine);
    }
}
}
```

通过调用 `getOptimalHeight()` 方法设置 `preferredHeight` 属性后, `layoutColumns()` 方法会循环访问各个 `TextField` 对象, 将每个对象的高度设置为 `preferredHeight` 值。然后, `layoutColumns()` 方法会为每个字段分布刚好足够的文本行, 以使每个字段中都不会发生滚动, 并且每个连续字段中的文本均会从前一个字段的文本结束处开始。如果文本对齐样式已设置为 “justify”, 则会调用 `justifyLastLine()` 方法以对齐字段中最后一行文本。否则, 最后一行将被视为段落结束行, 而不予对齐。

使用 Flash 文本引擎

Flash 文本引擎 (FTE) 为文本度量、格式设置和双向文本的复杂控制提供较低级别的支持。它提供了改进的文本流和增强的语言支持。尽管可以使用 FTE 创建和管理简单的文本元素, 但设计 FTE 的主要目的在于为开发人员创建文本处理组件提供基础。因此, Flash 文本引擎采用了更高级的编程技术。若要显示简单的文本元素, 请参阅前面介绍 `TextField` 以及相关对象的用法的部分。

注: 从 Flash Player 10 和 Adobe AIR 1.5 开始提供 Flash 文本引擎。

创建和显示文本

组成 Flash 文本引擎的类可用于创建文本、设置文本格式以及控制文本。下面的类是通过 Flash 文本引擎创建和显示文本时所用的基本构造块:

- `TextElement/GraphicElement/GroupElement`: 包含 `TextBlock` 实例的内容
- `ElementFormat`: 指定 `TextBlock` 实例内容的格式设置属性
- `TextBlock`: 用于构建文本段落的工厂
- `TextLine`: 依据 `TextBlock` 创建的文本行

若要显示文本, 请基于 `String` 和 `ElementFormat` 对象 (指定格式设置特征) 创建一个 `TextElement` 对象, 然后将该 `TextElement` 分配给 `TextBlock` 对象的 `content` 属性。通过调用 `TextBlock.createTextLine()` 方法创建将要显示的文本行。例如, 下面的代码使用这些 FTE 类显示 “Hello World! This is Flash Text Engine!”, 显示时使用默认的格式和字体值。

```
package
{
    import flash.text.engine.*;
    import flash.display.Sprite;

    public class HelloWorldExample extends Sprite
    {
        public function HelloWorldExample()
        {
            var str = "Hello World! This is Flash Text Engine!";
            var format:ElementFormat = new ElementFormat();
            var textElement:TextElement = new TextElement(str, format);
            var textBlock:TextBlock = new TextBlock();
            textBlock.content = textElement;

            var textLine1:TextLine = textBlock.createTextLine(null, 300);
            addChild(textLine1);
            textLine1.x = 30;
            textLine1.y = 30;
        }
    }
}
```

`createTextLine()` 的参数指定新行从哪一行开始，以及新行的宽度（单位为像素）。新行通常从前一行开始，但如果新行是第一行，则为 `null`。

添加 GraphicElement 和 GroupElement 对象

可以将 `GraphicElement` 对象分配给 `TextBlock` 对象以显示图像或图形元素。直接从图形或图像创建一个 `GraphicElement` 类的实例，然后将该实例分配给 `TextBlock.content` 属性。按照常规的方法调用 `TextBlock.createTextline()` 来创建文本行。下面的示例创建两个文本行，一个用 `GraphicElement` 对象创建，另一个用 `TextElement` 对象创建。

```
package
{
    import flash.text.engine.*;
    import flash.display.Sprite;
    import flash.display.Shape;
    import flash.display.Graphics;

    public class GraphicElementExample extends Sprite
    {
        public function GraphicElementExample()
        {
            var str:String = "Beware of Dog!";

            var triangle:Shape = new Shape();
            triangle.graphics.beginFill(0xFF0000, 1);
            triangle.graphics.lineStyle(3);
            triangle.graphics.moveTo(30, 0);
            triangle.graphics.lineTo(60, 50);
            triangle.graphics.lineTo(0, 50);
            triangle.graphics.lineTo(30, 0);
            triangle.graphics.endFill();

            var format:ElementFormat = new ElementFormat();
```

```
format.fontSize = 20;

var graphicElement:GraphicElement = new GraphicElement(triangle, triangle.width,
triangle.height, format);
var textBlock:TextBlock = new TextBlock();
textBlock.content = graphicElement;
var textLine1:TextLine = textBlock.createTextLine(null, triangle.width);
textLine1.x = 50;
textLine1.y = 110;
addChild(textLine1);

var textElement:TextElement = new TextElement(str, format);
textBlock.content = textElement;
var textLine2 = textBlock.createTextLine(null, 300);
addChild(textLine2);
textLine2.x = textLine1.x - 30;
textLine2.y = textLine1.y + 15;
}

}

}
```

可以创建一个 `GroupElement` 对象来创建一组 `TextElement`、`GraphicElement` 以及其它 `GroupElement` 对象，以将这些对象分配给 `TextBlock` 对象的 `content` 属性。`GroupElement()` 构造函数的参数是一个 `Vector`，它指向构成组的文本、图形以及组元素。下面的示例将两个图形元素和一个文本元素组成一组，然后将它们作为一个单元分配给某个文本块。

```
package
{
    import flash.text.engine.*;
    import flash.display.Sprite;
    import flash.display.Shape;
    import flash.display.Graphics;

    public class GroupElementExample extends Sprite
    {
        public function GroupElementExample()
        {
            var str:String = "Beware of Alligators!";

            var triangle1:Shape = new Shape();
            triangle1.graphics.beginFill(0xFF0000, 1);
            triangle1.graphics.lineStyle(3);
            triangle1.graphics.moveTo(30, 0);
            triangle1.graphics.lineTo(60, 50);
            triangle1.graphics.lineTo(0, 50);
            triangle1.graphics.lineTo(30, 0);
            triangle1.graphics.endFill();

            var triangle2:Shape = new Shape();
            triangle2.graphics.beginFill(0xFF0000, 1);
            triangle2.graphics.lineStyle(3);
            triangle2.graphics.moveTo(30, 0);
            triangle2.graphics.lineTo(60, 50);
            triangle2.graphics.lineTo(0, 50);
            triangle2.graphics.lineTo(30, 0);
            triangle2.graphics.endFill();
        }
    }
}
```

```
        var format:ElementFormat = new ElementFormat();
        format.fontSize = 20;
        var graphicElement1:GraphicElement = new GraphicElement(triangle1, triangle1.width,
triangle1.height, format);
        var textElement:TextElement = new TextElement(str, format);
        var graphicElement2:GraphicElement = new GraphicElement(triangle2, triangle2.width,
triangle2.height, format);
        var groupVector:Vector.<ContentElement> = new Vector.<ContentElement>();
        groupVector.push(graphicElement1, textElement, graphicElement2);
        var groupElement = new GroupElement(groupVector);
        var textBlock:TextBlock = new TextBlock();
        textBlock.content = groupElement;
        var textLine:TextLine = textBlock.createTextLine(null, 800);
        addChild(textLine);
        textLine.x = 100;
        textLine.y = 200;
    }
}
}
```

替换文本

您可以调用 `TextElement.replaceText()` 来替换您分配给 `TextBlock.content` 属性的 `TextElement` 中的文本，从而替换 `TextBlock` 实例中的文本。下面的示例使用 `repasteText()` 将文本插在行首、追加在行尾以及替换行中的文本。

```
package
{
    import flash.text.engine.*;
    import flash.display.Sprite;

    public class ReplaceTextExample extends Sprite
    {
        public function ReplaceTextExample()
        {

            var str:String = "Lorem ipsum dolor sit amet";
            var fontDescription:FontDescription = new FontDescription("Arial");
            var format:ElementFormat = new ElementFormat(fontDescription);
            format.fontSize = 14;
            var textElement:TextElement = new TextElement(str, format);
            var textBlock:TextBlock = new TextBlock();
            textBlock.content = textElement;
            createLine(textBlock, 10);
            textElement.replaceText(0, 0, "A text fragment: ");
            createLine(textBlock, 30);
            textElement.replaceText(43, 43, "...");
            createLine(textBlock, 50);
            textElement.replaceText(23, 28, "(ipsum)");
            createLine(textBlock, 70);
        }

        function createLine(textBlock:TextBlock, y:Number):void {
            var textLine:TextLine = textBlock.createTextLine(null, 300);
            textLine.x = 10;
            textLine.y = y;
            addChild(textLine);
        }
    }
}
```

该 `replaceText()` 方法将 `beginIndex` 和 `endIndex` 参数指定的文本替换为 `newText` 参数指定的文本。如果 `beginIndex` 和 `endIndex` 参数的值相同，`replaceText()` 将在相应的位置插入指定的文本。否则，将用新文本替换 `beginIndex` 和 `endIndex` 所指定的字符。

处理事件 在 FTE 中

您可以将事件侦听器添加到 `TextLine` 实例中，就像添加到其它显示对象一样。例如，您可以检测用户何时将鼠标移动到某个文本行的上方，或者用户何时单击此行。下面的示例检测这两种事件。如果将鼠标移动到此行的上方，光标将变为按钮光标，并且如果单击此行，此行将会变色。

```
package
{
    import flash.text.engine.*;
    import flash.ui.Mouse;
    import flash.display.Sprite
    import flash.events.MouseEvent;
    import flash.events.EventDispatcher;

    public class EventHandlerExample extends Sprite
    {
        var textBlock:TextBlock = new TextBlock();

        public function EventHandlerExample():void
        {
            var str:String = "I'll change color if you click me.";
            var fontDescription:FontDescription = new FontDescription("Arial");
            var format:ElementFormat = new ElementFormat(fontDescription, 18);
            var textElement = new TextElement(str, format);
            textBlock.content = textElement;
            createLine(textBlock);
        }

        private function createLine(textBlock:TextBlock):void
        {
            var textLine:TextLine = textBlock.createTextLine(null, 500);
            textLine.x = 30;
            textLine.y = 30;
            addChild(textLine);
            textLine.addEventListener("mouseOut", mouseOutHandler);
            textLine.addEventListener("mouseOver", mouseOverHandler);
            textLine.addEventListener("click", clickHandler);
        }

        private function mouseOverHandler(event:MouseEvent):void
        {
            Mouse.cursor = "button";
        }

        private function mouseOutHandler(event:MouseEvent):void
        {
            Mouse.cursor = "arrow";
        }

        function clickHandler(event:MouseEvent):void {
            if(textBlock.firstLine)
                removeChild(textBlock.firstLine);
            var newFormat:ElementFormat = textBlock.content.elementFormat.clone();
            newFormat.bold = true;
            textBlock.content.elementFormat = newFormat;
        }
    }
}
```

```
switch(newFormat.color)
{
    case 0x000000:
        newFormat.color = 0xFF0000;
        break;
    case 0xFF0000:
        newFormat.color = 0x00FF00;
        break;
    case 0x00FF00:
        newFormat.color = 0x0000FF;
        break;
    case 0x0000FF:
        newFormat.color = 0x000000;
        break;
}
textBlock.content.elementFormat = newFormat;
createLine(textBlock);
}
}
```

镜像事件

另外，还可将文本块或文本块局部的事件映射到事件调度程序。首先，创建一个 `EventDispatcher` 实例，然后将其分配给 `TextElement` 实例的 `eventMirror` 属性。如果文本块由一个文本元素组成，那么文本引擎将映射整个文本块的事件。如果文本块由多个文本元素组成，那么文本引擎只映射设置了 `eventMirror` 属性的 `TextElement` 实例的事件。下面示例中的文本由三个元素组成：单词“Click”、单词“here”以及字符串“to see me in italic”。此示例将事件调度程序分配给第二个文本元素，即单词“here”，然后添加事件侦听器 `clickHandler()` 方法。该 `clickHandler()` 方法将文本更改为斜体。另外，它还将第三个文本元素的内容替换为“Click here to see me in normal font!”。

```
package
{
    import flash.text.engine.*;
    import flash.ui.Mouse;
    import flash.display.Sprite;
    import flash.events.MouseEvent;
    import flash.events.EventDispatcher;

    public class EventMirrorExample extends Sprite
    {
        var fontDescription:FontDescription = new FontDescription("Helvetica", "bold");
        var format:ElementFormat = new ElementFormat(fontDescription, 18);
        var textElement1 = new TextElement("Click ", format);
        var textElement2 = new TextElement("here ", format);
        var textElement3 = new TextElement("to see me in italic! ", format);
        var textBlock:TextBlock = new TextBlock();

        public function EventMirrorExample()
        {
            var myEvent:EventDispatcher = new EventDispatcher();

            myEvent.addEventListener("click", clickHandler);
            myEvent.addEventListener("mouseOut", mouseOutHandler);
            myEvent.addEventListener("mouseOver", mouseOverHandler);

            textElement2.eventMirror=myEvent;

            var groupVector:Vector.<ContentElement> = new Vector.<ContentElement>;
            groupVector.push(textElement1, textElement2, textElement3);
            var groupElement:GroupElement = new GroupElement(groupVector);
        }
    }
}
```

```
textBlock.content = groupElement;
createLines(textBlock);
}

private function clickHandler(event:MouseEvent):void
{
    var newFont:FontDescription = new FontDescription();
    newFont.fontWeight = "bold";

    var newFormat:ElementFormat = new ElementFormat();
    newFormat.fontSize = 18;
    if(textElement3.text == "to see me in italic! ") {
        newFont.fontPosture = FontPosture.ITALIC;
        textElement3.replaceText(0,21, "to see me in normal font! ");
    }
    else {
        newFont.fontPosture = FontPosture.NORMAL;
        textElement3.replaceText(0, 26, "to see me in italic! ");
    }
    newFormat.fontDescription = newFont;
    textElement1.elementFormat = newFormat;
    textElement2.elementFormat = newFormat;
    textElement3.elementFormat = newFormat;
    createLines(textBlock);
}

private function mouseOverHandler(event:MouseEvent):void
{
    Mouse.cursor = "button";
}

private function mouseOutHandler(event:MouseEvent):void
{
    Mouse.cursor = "arrow";
}

private function createLines(textBlock:TextBlock):void
{
    if(textBlock.firstLine)
        removeChild (textBlock.firstLine);
    var textLine:TextLine = textBlock.createTextLine (null, 300);
    textLine.x = 15;
    textLine.y = 20;
    addChild (textLine);
}
}
```

mouseOverHandler() 和 mouseOutHandler() 函数在光标位于单词 “here” 上方时将光标设置为按钮光标，在光标不在 “here” 上方时将光标恢复为箭头。

设置文本格式

TextBlock 对象是用于创建文本行的工厂。TextBlock 的内容是通过 TextElement 对象分配的。ElementFormat 对象负责处理文本的格式设置。ElementFormat 类定义基线对齐、字距调整、间距、文本旋转以及字体大小、颜色和大小写等属性。它还包含 FontDescription，相关内容在第 390 页的“[使用字体](#)”中详细介绍。

使用 ElementFormat 对象

ElementFormat 对象的构造函数可以接受一长串可选参数中的任何一个参数，包括 FontDescription。也可以在构造函数之外设置这些属性。下面的示例演示了在定义和显示简单文本行时各个对象的关系：

```
package
{
    import flash.display.Sprite;
    import flash.text.*;

    public class ElementFormatExample extends Sprite
    {
        private var tb:TextBlock = new TextBlock();
        private var te:TextElement;
        private var ef:ElementFormat;
        private var fd:FontDescription = new FontDescription();
        private var str:String;
        private var tl:TextLine;

        public function ElementFormatExample()
        {
            fd.fontName = "Garamond";
            ef = new ElementFormat(fd);
            ef.fontSize = 30;
            ef.color = 0xFF0000;
            str = "This is flash text";
            te = new TextElement(str, ef);
            tb.content = te;
            tl = tb.createTextLine(null, 600);
            addChild(tl);
        }
    }
}
```

字体颜色和透明度 (alpha)

ElementFormat 对象的 color 属性设置字体颜色。该值是一个表示 RGB 颜色分量的整数；例如，0xFF0000 表示红色，0x00FF00 表示绿色。默认值是黑色 (0x000000)。

alpha 属性设置元素 (TextElement 和 GraphicElement) 的 alpha 透明度值。值范围介于 0 (完全透明) 到 1 (完全不透明，**默认值**) 之间。alpha 为 0 的元素不可见，但仍为活动元素。此值乘以继承的 alpha 值，因而元素更为透明。

```
var ef:ElementFormat = new ElementFormat();
ef.alpha = 0.8;
ef.color = 0x999999;
```

基线对齐和移位

行中最大文本的字体和大小决定了行的主要基线。可通过设置 TextBlock.baselineFontDescription 和 TextBlock.baselineFontSize 来覆盖这些值。可以将主要基线与文本内多个基线中的一个对齐。这些基线包括上升线和下降线或表意顶部、中心或底部。



A. 上升 B. 基线 C. 下降 D. x 高度

在 ElementFormat 对象中，三个属性决定了基线和对齐特征。alignmentBaseline 属性设置 TextElement 或 GraphicElement 的主基线。此基线是元素的“贴紧”线，所有文本的主要基线均对齐此位置。

dominantBaseline 属性指定要使用各个元素基线中的哪个基线，这决定了元素在行中的垂直位置。默认值为 TextBaseline.ROMAN，但也可以设置为使用 IDEOGRAPHIC_TOP 基线或 IDEOGRAPHIC_BOTTOM 基线作为主要基线。

baselineShift 属性将基线沿 y 轴移动一定数量的像素。在正常（非旋转）文本中，正数值使基线下移，负数值使基线上移。

印刷大小写

ElementFormat 的 TypographicCase 属性指定文本大小写，如大写、小写或小型大写字母。

```
var ef_Upper:ElementFormat = new ElementFormat();
ef_Upper.typographicCase = TypographicCase.UPPERCASE;

var ef_SmallCaps:ElementFormat = new ElementFormat();
ef_SmallCaps.typographicCase = TypographicCase.SMALL_CAPS;
```

旋转文本

可以 90 度为增量旋转文本块或文本段中的字型。TextRotation 类定义了以下常量，用于设置文本块和字型旋转：

常数	值	描述
AUTO	“auto”	指定 90 度逆时针旋转。通常用于垂直的亚洲文字，以仅旋转需要旋转的字型。
ROTATE_0	“rotate_0”	指定不进行旋转。
ROTATE_180	“rotate_180”	指定 180 度旋转。
ROTATE_270	“rotate_270”	指定 270 度旋转。
ROTATE_90	“rotate_90”	指定 90 度顺时针旋转。

若要旋转文本块中的多行文本，请在调用 TextBlock.createTextLine() 方法创建文本行之前设置 TextBlock.lineRotation 属性。

若要旋转文本块或文本段中的字型，请将 ElementFormat.textRotation 属性设置为字型要旋转的度数。字型是构成字符的形状，或者是由多个字型组成的一部分字符。例如，字母 a 和 i 上的点都是字型。

在某些亚洲语言中会涉及旋转字型。在这些语言中，需要将行旋转到垂直方向，但不旋转行中的字符。有关旋转亚洲文字的详细信息，请参阅第 393 页的“[对齐东亚文字](#)”。

下面的示例同时旋转文本块以及其中的字型，就像处理亚洲文字那样。该示例也使用日文字体：

```
package
{
    import flash.display.Sprite;
    import flash.text.*;

    public class RotationExample extends Sprite
    {
        private var tb:TextBlock = new TextBlock();
        private var te:TextElement;
        private var ef:ElementFormat;
        private var fd:FontDescription = new FontDescription();
        private var str:String;
        private var tl:TextLine;

        public function RotationExample()
        {
            fd.fontName = "MS Mincho";
            ef = new ElementFormat(fd);
            ef.textRotation = TextRotation.AUTO;
            str = "This is rotated Japanese text";
            te = new TextElement(str, ef);
            tb.lineRotation = TextRotation.ROTATE_90;
            tb.content = te;
            tl = tb.createTextLine(null, 600);
            addChild(tl);
        }
    }
}
```

锁定和克隆 ElementFormat

当 ElementFormat 对象分配给任何类型的 ContentElement 时，其 locked 属性自动设置为 true。试图修改锁定的 ElementFormat 对象将会引发 IllegalOperationError。最佳做法是先完全定义这类对象，然后将其分配给 TextElement 实例。

如果要修改某个现有 ElementFormat 实例，请首先检查其 locked 属性。如果该属性为 true，请使用 clone() 方法创建对象的未锁定副本。此未锁定对象的属性可以更改，然后可以将其指派给 TextElement 实例。依据此对象创建的所有新行均具有新的格式设置。前面依据此对象创建并使用旧格式的行保持不变。

```

package
{
    import flash.display.Sprite;
    import flash.text.*;

    public class ElementFormatCloneExample extends Sprite
    {
        private var tb:TextBlock = new TextBlock();
        private var te:TextElement;
        private var ef1:ElementFormat;
        private var ef2:ElementFormat;
        private var fd:FontDescription = new FontDescription();

        public function ElementFormatCloneExample()
        {
            fd.fontName = "Garamond";
            ef1 = new ElementFormat(fd);
            ef1.fontSize = 24;
            var str:String = "This is flash text";
            te = new TextElement(str, ef);
            tb.content = te;
            var tx1:TextLine = tb.createTextLine(null, 600);
            addChild(tx1);

            ef2 = (ef1.locked) ? ef1.clone() : ef1;
            ef2.fontSize = 32;
            tb.content.elementFormat = ef2;
            var tx2:TextLine = tb.createTextLine(null, 600);
            addChild(tx2);
        }
    }
}

```

使用字体

FontDescription 对象与 ElementFormat 配合使用可以标识字型，以及定义其部分特征。这些特征包括字体名称、粗细、形态、呈现以及如何查找字体（设备字体与嵌入字体）。

注：FTE 不支持 Type 1 字体或 Type 3、ATC、sfnt 换行 CID 或 Naked CID 等位图字体。

定义字体特征（FontDescription 对象）

FontDescription 对象的 fontName 属性可以是单个名称，也可以是逗号分隔的名称列表。例如，在诸如“Arial, Helvetica, _sans”的列表中，Flash Player 或 AIR 首先查找“Arial”，然后查找“Helvetica”，如果这两种字体都找不到，则最后寻找“_sans”。在字体名称集中有三个通用设备字体名称：“_sans”、“_serif”和“_typewriter”。这些名称根据播放系统映射到具体的设备字体。在使用设备字体的所有字体说明中指定此类默认名称是一种很好的做法。如果没有指定任何 fontName，则默认使用“_serif”。

fontPosture 属性可设置为默认值 (FontPosture.NORMAL)，也可设置为斜体 (FontPosture.ITALIC)。fontWeight 属性可设置为默认值 (FontWeight.NORMAL)，也可设置为粗体 (FontWeight.BOLD)。

```

var fd1:FontDescription = new FontDescription();
fd1.fontName = "Arial, Helvetica, _sans";
fd1.fontPosture = FontPosture.NORMAL;
fd1.fontWeight = FontWeight.BOLD;

```

嵌入字体和设备字体

FontDescription 对象的 fontLookup 属性指定 Flash Player 和 AIR 是查找设备字体还是查找嵌入字体来呈现文本。如果指定设备字体 (FontLookup.DEVICE)，则运行时将在回放系统中查找该字体。如果指定嵌入字体 (FontLookup.EMBEDDED_CFF)，运行时将在 SWF 文件中查找具有指定名称的嵌入字体。只有嵌入的 CFF (压缩字体格式) 字体才可使用此项设置。如果没有找到指定的字体，则会使用后备设备字体。

使用设备字体时，得到的 SWF 文件较小。嵌入字体提供了更高的跨平台保真度。

```
var fd1:FontDescription = new FontDescription();
fd1.fontLookup = FontLookup.EMBEDDED_CFF;
fd1.fontName = "Garamond, _serif";
```

呈现模式和提示

从 Flash Player 10 和 Adobe AIR 1.5 开始提供 CFF (压缩字体格式) 呈现。以此方式呈现字体能使文本更加清晰可辨，并为小字号的字体提供更高品质的显示效果。此项设置只适用于嵌入字体。对于 renderingMode 属性，FontDescription 默认设为此项设置 (RenderingMode.CFF)。可以将此属性设置为 RenderingMode.NORMAL，从而与 Flash Player 7 或更低版本所使用的呈现类型保持一致。

选择 CFF 呈现之后，将由另一个属性 cffHinting 控制字体水平线适合子像素网格的方式。默认值 CFFHinting.HORIZONTAL_STEM 使用 CFF 提示。将此属性设置为 CFFHinting.NONE 将会消除提示，对于动画或大字号应当如此设置。

```
var fd1:FontDescription = new FontDescription();
fd1.renderingMode = RenderingMode.CFF;
fd1.cffHinting = CFFHinting.HORIZONTAL_STEM;
```

锁定和克隆 FontDescription

当 FontDescription 对象分配给 ElementFormat 时，其 locked 属性自动设置为 true。试图修改锁定的 FontDescription 对象将会引发 IllegalOperationError。最佳做法是先完全定义这类对象，然后将其分配给 ElementFormat。

如果要修改某个现有 FontDescription，请首先检查其 locked 属性。如果该属性为 true，请使用 clone() 方法创建对象的未锁定副本。此未锁定对象的属性可以更改，然后可以将其指派给 ElementFormat。根据此 TextElement 创建的所有新行均具有新的格式设置。以前根据这个对象创建的行保持不变。

```
package
{
    import flash.display.Sprite;
    import flash.text.*;

    public class FontDescriptionCloneExample extends Sprite
    {
        private var tb:TextBlock = new TextBlock();
        private var te:TextElement;
        private var ef1:ElementFormat;
        private var ef2:ElementFormat;
        private var fd1:FontDescription = new FontDescription();
        private var fd2:FontDescription;

        public function FontDescriptionCloneExample()
        {
            fd1.fontName = "Garamond";
            ef1 = new ElementFormat(fd1);
            var str:String = "This is flash text";
            te = new TextElement(str, ef1);
            tb.content = te;
            var tx1:TextLine = tb.createTextLine(null, 600);
            addChild(tx1);

            fd2 = (fd1.locked) ? fd1.clone() : fd1;
            fd2.fontName = "Arial";
            ef2 = (ef1.locked) ? ef1.clone() : ef1;
            ef2.fontDescription = fd2;
            tb.content.elementFormat = ef2;
            var tx2:TextLine = tb.createTextLine(null, 600);
            addChild(tx2);
        }
    }
}
```

控制文本

FTE 提供了一组新的文本格式控件，用以处理对齐和字符间距（字距调整和间距）。还有一些属性可用于控制各行断开方式以及在行中设置 Tab 停靠位。

对齐文本

通过调整单词之间（有的时候是字母之间）的间距，对齐文本能让段落中各行的长度相等。其效果是文本两端对齐，而单词和字母之间的间距发生变化。报刊杂志中的文本栏通常是对齐的。

SpaceJustifier 类中的 **lineJustification** 属性可用于控制文本块中各行的对齐情况。**LineJustification** 类定义了可用于指定对齐选项的常量：**ALL_BUT_LAST** 对齐除文本最后一行之外的所有各行；**ALL_INCLUDING_LAST** 对齐所有文本，包括最后一行；**UNJUSTIFIED** 是默认值，保持文本的不对齐状态。

若要对齐文本，请将 **lineJustification** 属性设置为 **SpaceJustifier** 类的实例，并将该实例指派给 **TextBlock** 实例的 **textJustifier** 属性。下面的示例创建一个段落。在此段落中，除最后一行文本之外，所有各行均对齐。

```

package
{
    import flash.text.engine.*;
    import flash.display.Sprite;

    public class JustifyExample extends Sprite
    {
        public function JustifyExample()
        {
            var str:String = "Lorem ipsum dolor sit amet, consectetur adipisicing elit, " +
                "sed do eiusmod tempor incididunt ut labore et dolore magna aliqua. Ut " +
                "enim ad minim veniam, quis nostrud exercitation ullamco laboris nisi ut " +
                "aliquip ex ea commodo consequat.";

            var format:ElementFormat = new ElementFormat();
            var textElement:TextElement=new TextElement(str,format);
            var spaceJustifier:SpaceJustifier=new SpaceJustifier("en",LineJustification.ALL_BUT_LAST);

            var textBlock:TextBlock = new TextBlock();
            textBlock.content=textElement;
            textBlock.textJustifier=spaceJustifier;
            createLines(textBlock);
        }

        private function createLines(textBlock:TextBlock):void {
            var yPos=20;
            var textLine:TextLine=textBlock.createTextLine(null,150);

            while (textLine) {
                addChild(textLine);
                textLine.x=15;
                yPos+=textLine.textHeight+2;
                textLine.y=yPos;
                textLine=textBlock.createTextLine(textLine,150);
            }
        }
    }
}

```

若要更改字母以及单词之间的间距，请将 `SpaceJustifier.letterspacing` 属性设置为 `true`。应用字母间距调整功能可以减少单词间出现不美观间隙（简单对齐有时会出现这种情况）的几率。

对齐东亚文字

对齐东亚文字时，需要额外注意一些事项。东亚文字可能是从上至下书写，且某些字符（称为避头尾）不能出现在行首或行尾。`JustificationStyle` 类定义了以下常量，这些常量指定了用于处理这些字符的选项。`PRIORITYIZE_LEAST_ADJUSTMENT` 通过扩展现行或压缩行进行对齐，具体采用哪种方式取决于能否产生最理想的效果。`PUSH_IN_KINSOKU` 通过行尾压缩避头尾或扩展现行（如果没有避头尾或该空间不足）进行对齐。

`PUSH_OUT_ONLY` 通过扩展现行进行对齐。若要创建垂直的亚洲文本块，请将 `TextBlock.lineRotation` 属性设置为 `TextRotation.ROTATE_90`，并将 `ElementFormat.textRotation` 属性设置为 `TextRotation.AUTO`（默认值）。将 `textRotation` 属性设置为 `AUTO` 将使文本中的字型保持垂直，而不是在整行发生旋转的时候侧向旋转。`AUTO` 设置根据字型的 `Unicode` 属性，只逆时针旋转全角字型和宽字型 90 度。下面的示例显示垂直的日文文本块，并使用 `PUSH_IN_KINSOKU` 选项使其对齐。

i

```

package
{
    import flash.text.engine.*;
    import flash.display.Stage;
    import flash.display.Sprite;
    import flash.system.Capabilities;

    public class EastAsianJustifyExample extends Sprite
    {
        public function EastAsianJustifyExample()
        {
            var Japanese_txt:String = String.fromCharCode(
                0x5185, 0x95A3, 0x5E9C, 0x304C, 0x300C, 0x653F, 0x5E9C, 0x30A4,
                0x30F3, 0x30BF, 0x30FC, 0x30CD, 0x30C3, 0x30C8, 0x30C6, 0x30EC,
                0x30D3, 0x300D, 0x306E, 0x52D5, 0x753B, 0x914D, 0x4FE1, 0x5411,
                0x3051, 0x306B, 0x30A2, 0x30C9, 0x30D3, 0x30B7, 0x30B9, 0x30C6,
                0x30E0, 0x30BA, 0x793B, 0x306E)
            var textBlock:TextBlock = new TextBlock();
            var font:FontDescription = new FontDescription();
            var format:ElementFormat = new ElementFormat();
            format.fontSize = 12;
            format.color = 0xCC0000;
            format.textRotation = TextRotation.AUTO;
            textBlock.baselineZero = TextBaseline.IDEOGRAPHIC_CENTER;
            var eastAsianJustifier:EastAsianJustifier = new EastAsianJustifier("ja",
LineJustification.ALL_BUT_LAST);
            eastAsianJustifier.justificationStyle = JustificationStyle.PUSH_IN_KINSOKU;
            textBlock.textJustifier = eastAsianJustifier;
            textBlock.lineRotation = TextRotation.ROTATE_90;
            var linePosition:Number = this.stage.stageWidth - 75;
            if (Capabilities.os.search("Mac OS") > -1)
                // set fontName: Kozuka Mincho Pro R
                font.fontName = String.fromCharCode(0x5C0F, 0x585A, 0x660E, 0x671D) + " Pro R";
            else
                font.fontName = "Kozuka Mincho Pro R";
            textBlock.content = new TextElement(Japanese_txt, format);
            var previousLine:TextLine = null;


            while (true)
            {
                var textLine:TextLine = textBlock.createTextLine(previousLine, 200);
                if (textLine == null)
                    break;
                textLine.y = 20;
                textLine.x = linePosition;
                linePosition -= 25;
                addChild(textLine);
                previousLine = textLine;
            }
        }
    }
}

```

字距调整和间距

字距调整和间距影响文本块中相邻两个字符之间的距离。字距调整控制字符对如何组合在一起，例如字符对“WA”或“Va”。字距调整在 ElementFormat 对象中设置。字距调整默认为启用 (Kerning.ON)。它可以设置为 OFF 或 AUTO，在此情形中，只有两个字符都不是日本汉字、平假名和片假名时，才在它们之间应用字距调整。

间距处理在文本块中的所有字符之间增加或减少一定数量的像素，并且也是在 ElementFormat 对象中进行设置。嵌入字体和设备字体均可使用间距处理。FTE 支持两种间距属性：trackingLeft，增加 / 减少字符左侧的像素；以及 trackingRight，增加 / 减少字符右侧的像素。如果使用间距，那么每个字符对的字距调整值将会加上或减去间距值。



A. Kerning.OFF B. TrackingRight=5, Kerning.OFF C. TrackingRight=-5, Kerning.OFF D. Kerning.ON E. TrackingRight=-5, Kerning.ON F. TrackingRight=-5, Kerning.ON

```
var ef1:ElementFormat = new ElementFormat();
ef1.kerning = Kerning.OFF;

var ef2:ElementFormat = new ElementFormat();
ef2.kerning = Kerning.ON;
ef2.trackingLeft = 0.8;
ef2.trackingRight = 0.8;

var ef3:ElementFormat = new ElementFormat();
ef3.trackingRight = -0.2;
```

换行文本的换行符

ElementFormat 对象的 breakOpportunity 属性确定在换行文本分成多行时，哪些字符可用于断行。默认值 BreakOpportunity.AUTO 使用标准 Unicode 属性，例如在不同单词之间断行，以及在连字符的位置断行。如果使用 BreakOpportunity.ALL，那么在任何一个字符均可断行，这有助于制造某些效果（例如沿某个路径排列文本）。

```
var ef:ElementFormat = new ElementFormat();
ef.breakOpportunity = BreakOpportunity.ALL;
```

Tab 停靠位

若要在文本块中设置 Tab 停靠位，请通过创建 TabStop 类的实例来定义 Tab 停靠位。TabStop() 构造函数的参数指定文本如何与 Tab 停靠位对齐。这些参数指定 Tab 停靠位的位置，以及对于十进制数对齐，指定要在哪个值对齐（表示为字符串）。通常，此值是小数点，但也可以是逗号、美元符号或者日元或欧元符号等等。下面的代码行创建一个名为 tab1 的 Tab 停靠位。

```
var tab1:TabStop = new TabStop(TabIndex.Decimal, 50, ".");
```

为一个文本块创建 Tab 停靠位之后，请将这些停靠位分配给 TextBlock 实例的 tabStops 属性。由于 tabStops 属性需要一个 Vector，首先创建一个 Vector，然后将 Tab 停靠位添加到此 Vector 中。此 Vector 允许您将一组 Tab 停靠位分配给文本块。下面的示例创建一个 Vector<TabStop> 实例，并将一组 TabStop 对象添加到其中。然后，代码将这些 Tab 停靠位分配给 TextBlock 实例的 tabStops 属性。

```
var tabStops:Vector.<TabStop> = new Vector.<TabStop>();
tabStops.push(tab1, tab2, tab3, tab4);
textBlock.tabStops = tabStops
```

有关矢量的详细信息，请参阅第 128 页的“[处理数组](#)”。

下面的示例演示每个 TabStop 对齐选项的效果。

```

package {

    import flash.text.engine.*;
    import flash.display.Sprite;

    public class TabStopExample extends Sprite
    {
        public function TabStopExample()
        {
            var format:ElementFormat = new ElementFormat();
            format.fontDescription = new FontDescription("Arial");
            format.fontSize = 16;

            var tabStops:Vector.<TabStop> = new Vector.<TabStop>();
            tabStops.push(
                new TabStop(TabAlignment.START, 20),
                new TabStop(TabAlignment.CENTER, 140),
                new TabStop(TabAlignment.DECIMAL, 260, "."),
                new TabStop(TabAlignment.END, 380));
            var textBlock:TextBlock = new TextBlock();
            textBlock.content = new TextElement(
                "\tt1\tt2\tt3\tt4\n" +
                "\tThis line aligns on 1st tab\n" +
                "\t\t\t\tThis is the end\n" +
                "\tThe following fragment centers on the 2nd tab:\t\t\n" +
                "\t\tit's on me\t\t\n" +
                "\tThe following amounts align on the decimal point:\n" +
                "\t\t\t45.00\t\n" +
                "\t\t\t75,320.00\t\n" +
                "\t\t\t6,950.00\t\n" +
                "\t\t\t7.01\t", format);

            textBlock.tabStops = tabStops;
            var yPosition:Number = 60;
            var previousTextLine:TextLine = null;
            var textLine:TextLine;
            var i:int;
            for (i = 0; i < 10; i++) {
                textLine = textBlock.createTextLine(previousTextLine, 1000, 0);
                textLine.x = 20;
                textLine.y = yPosition;
                addChild(textLine);
                yPosition += 25;
                previousTextLine = textLine;
            }
        }
    }
}

```

FTE 示例 - 新闻布局

此编程示例演示在为简单的新闻页面布局时，如何使用 Flash 文本引擎。该页包含大标题、副标题和多列的正文部分。

首先创建一个新的 FLA 文件，并将下面的代码附加到默认图层的第 2 帧：

```

import com.example.programmingas3.newslayout.StoryLayout ;
// frame script - create a 3-columnmed article layout
var story:StoryLayout = new StoryLayout(720, 500, 3, 10);
story.x = 20;
story.y = 80;
addChild(story);
stop();

```

StoryLayout.as 是本示例的控制器脚本。它设置内容，从外部样式表中读取样式信息，并将这些样式指派给 **ElementFormat** 对象。然后，它将创建标题、副标题和多列文本元素。

```
package com.example.programmingas3.newslayout
{
    import flash.display.Sprite;
    import flash.text.StyleSheet;
    import flash.text.engine.*;

    import flash.events.Event;
    import flash.net.URLRequest;
    import flash.net.URLLoader;
    import flash.display.Sprite;
    import flash.display.Graphics;

    public class StoryLayout extends Sprite
    {
        public var headlineTxt:HeadlineTextField;
        public var subtitleTxt:HeadlineTextField;
        public var storyTxt:MultiColumnText;
        public var sheet:StyleSheet;
        public var h1_ElFormat:ElementFormat;
        public var h2_ElFormat:ElementFormat;
        public var p_ElFormat:ElementFormat;

        private var loader:URLLoader;

        public var paddingLeft:Number;
        public var paddingRight:Number;
        public var paddingTop:Number;
        public var paddingBottom:Number;

        public var preferredWidth:Number;
        public var preferredHeight:Number;

        public var numColumns:int;

        public var bgColor:Number = 0xFFFFFFFF;

        public var headline:String = "News Layout Example";
        public var subtitle:String = "This example formats text like a newspaper page using the Flash Text Engine API. ";

        public var rawTestData:String =
            "From the part Mr. Burke took in the American Revolution, it was natural that I should consider him a friend to mankind; and as our acquaintance commenced on that ground, it would have been more agreeable to me to have had cause to continue in that opinion than to change it. " +
            "At the time Mr. Burke made his violent speech last winter in the English Parliament against the French Revolution and the National Assembly, I was in Paris, and had written to him but a short time before to inform him how prosperously matters were going on. Soon after this I saw his advertisement of the Pamphlet he intended to publish: As the attack was to be made in a language but little studied, and less understood in France, and as everything suffers by translation, I promised some of the friends of the Revolution in that country that whenever Mr. Burke's Pamphlet came forth, I would answer it. This appeared to me the more necessary to be done, when I saw the flagrant misrepresentations which Mr. Burke's Pamphlet contains; and that while it is an outrageous abuse on the French Revolution, and the principles of Liberty, it is an imposition on the rest of the world. " +
            "I am the more astonished and disappointed at this conduct in Mr. Burke, as (from the circumstances I am going to mention) I had formed other expectations. " +
            "I had seen enough of the miseries of war, to wish it might never more have existence in the world, and that some other mode might be found out to settle the differences that should occasionally arise in the neighbourhood of nations. This certainly might be done if Courts were disposed to set honesty about it, or if countries were enlightened enough not to be made the dupes of Courts. The people of America had been bred
```

up in the same prejudices against France, which at that time characterised the people of England; but experience and an acquaintance with the French Nation have most effectually shown to the Americans the falsehood of those prejudices; and I do not believe that a more cordial and confidential intercourse exists between any two countries than between America and France. ";

```
public function StoryLayout(w:int = 400, h:int = 200, cols:int = 3, padding:int = 10):void
{
    this.preferredWidth = w;
    this.preferredHeight = h;

    this.numColumns = cols;

    this.paddingLeft = padding;
    this.paddingRight = padding;
    this.paddingTop = padding;
    this.paddingBottom = padding;

    var req:URLRequest = new URLRequest("story.css");
    loader = new URLLoader();
    loader.addEventListener(Event.COMPLETE, onCSSFileLoaded);
    loader.load(req);
}

public function onCSSFileLoaded(event:Event):void
{
    this.sheet = new StyleSheet();
    this.sheet.parseCSS(loader.data);

    // convert headline styles to ElementFormat objects
    h1_ElFormat = getElFormat("h1", this.sheet);
    h1_ElFormat.typographicCase = TypographicCase.UPPERCASE;
    h2_ElFormat = getElFormat("h2", this.sheet);
    p_ElFormat = getElFormat("p", this.sheet);
    displayText();
}

public function drawBackground():void
{
    var h:Number = this.storyTxt.y + this.storyTxt.height +
                  this.paddingTop + this.paddingBottom;
    var g:Graphics = this.graphics;
    g.beginFill(this.bgColor);
    g.drawRect(0, 0, this.width + this.paddingRight + this.paddingLeft, h);
    g.endFill();
}

/**
 * Reads a set of style properties for a named style and then creates
 * a TextFormat object that uses the same properties.
 */
public function getElFormat(styleName:String, ss:StyleSheet):ElementFormat
{
    var style:Object = ss.getStyle(styleName);
    if (style != null)
    {
        var colorStr:String = style.color;
        if (colorStr != null && colorStr.indexOf("#") == 0)
        {
            style.color = colorStr.substr(1);
        }
        var fd:FontDescription = new FontDescription(
            style.fontFamily,
```

```
        style.fontWeight,
        FontPosture.NORMAL,
        FontLookup.DEVICE,
        RenderingMode.NORMAL,
        CFFHinting.NONE);
var format:ElementFormat = new ElementFormat(fd,
        style.fontSize,
        style.color,
        1,
        TextRotation.AUTO,
        TextBaseline.ROMAN,
        TextBaseline.USE_DOMINANT_BASELINE,
        0.0,
        Kerning.ON,
        0.0,
        0.0,
        "en",
        BreakOpportunity.AUTO,
        DigitCase.DEFAULT,
        DigitWidth.DEFAULT,
        LigatureLevel.NONE,
        TypographicCase.DEFAULT);

if (style.hasOwnProperty("letterSpacing"))
{
    format.trackingRight = style.letterSpacing;
}
}
return format;
}

public function displayText():void
{
headlineTxt = new HeadlineTextField(h1_ElFormat,headline,this.preferredWidth);
headlineTxt.x = this.paddingLeft;
headlineTxt.y = 40 + this.paddingTop;
    headlineTxt.fitText(1);
    this.addChild(headlineTxt);

subtitleTxt = new HeadlineTextField(h2_ElFormat,subtitle,this.preferredWidth);
subtitleTxt.x = this.paddingLeft;
subtitleTxt.y = headlineTxt.y + headlineTxt.height;
    subtitleTxt.fitText(2);
this.addChild(subtitleTxt);

storyTxt = new MultiColumnText(rawTestData, this.numColumns,
        20, this.preferredWidth, this.preferredHeight, p_ElFormat);
storyTxt.x = this.paddingLeft;
storyTxt.y = subtitleTxt.y + subtitleTxt.height + 10;
this.addChild(storyTxt);

    drawBackground();
}
}
}
```

FormattedTextBlock.as 用作文本块创建的基类。它还包含用于更改字体大小和大小写的实用程序函数。

```
package com.example.programmingas3.newslayout
{
    import flash.text.engine.*;
    import flash.display.Sprite;

    public class FormattedTextBlock extends Sprite
    {
        public var tb:TextBlock;
        private var te:TextElement;
        private var efl:ElementFormat;
        private var textWidth:int;
        public var totalTextLines:int;
        public var blockText:String;
        public var leading:Number = 1.25;
        public var preferredWidth:Number = 720;
        public var preferredHeight:Number = 100;

        public function FormattedTextBlock(ef:ElementFormat,txt:String, colW:int = 0)
        {
            this.textWidth = (colW==0) ? preferredWidth : colW;
            blockText = txt;
            efl = ef;
            tb = new TextBlock();
            tb.textJustifier = new SpaceJustifier("en",LineJustification.UNJUSTIFIED,false);
            te = new TextElement(blockText,this.efl);
            tb.content = te;
            this.breakLines();
        }

        private function breakLines()
        {
            var textLine:TextLine = null;
            var y:Number = 0;
            var lineNum:int = 0;
            while (textLine = tb.createTextLine(textLine,this.textWidth,0,true))
            {
                textLine.x = 0;
                textLine.y = y;
                y += this.leading*textLine.height;
                this.addChild(textLine);
            }
            for (var i:int = 0; i < this.numChildren; i++)
            {
                TextLine(this.getChildAt(i)).validity = TextLineValidity.STATIC;
            }
            this.totalTextLines = this.numChildren;
        }

        private function rebreakLines()
        {
            this.clearLines();
            this.breakLines();
        }

        private function clearLines()
        {
            while(this.numChildren)
            {
                this.removeChildAt(0);
            }
        }
    }
}
```

```
public function changeSize(size:uint=12):void
{
    if (size > 5)
    {
        var ef2:ElementFormat = ef1.clone();
        ef2.fontSize = size;
        te.elementFormat = ef2;
        this.rebreakLines();
    }
}

public function changeCase(newCase:String = "default"):void
{
    var ef2:ElementFormat = ef1.clone();
    ef2.typographicCase = newCase;
    te.elementFormat = ef2;
}
}
```

HeadlineTextBlock.as 扩展 FormattedTextBlock 类，用于创建标题。它包含用于将文本放入已定义页面区域的函数。

```
package com.example.programmingas3.newslayout
{
    import flash.text.engine.*;
    public class HeadlineTextField extends FormattedTextBlock
    {

        public static var MIN_POINT_SIZE:uint = 6;
        public static var MAX_POINT_SIZE:uint = 128;

        public function HeadlineTextField(te:ElementFormat,txt:String,colW:int = 0)
        {
            super(te,txt);
        }

        public function fitText(maxLines:uint = 1, targetWidth:Number = -1):uint
        {
            if (targetWidth == -1)
            {
                targetWidth = this.width;
            }

            var pixelsPerChar:Number = targetWidth / this.blockText.length;
            var pointSize:Number = Math.min(MAX_POINT_SIZE,
                Math.round(pixelsPerChar * 1.8 * maxLines));

            if (pointSize < 6)
            {
                // the point size is too small
                return pointSize;
            }

            this.changeSize(pointSize);
            if (this.totalTextLines > maxLines)
            {
                return shrinkText(--pointSize, maxLines);
            }
            else
            {
                return growText(pointSize, maxLines);
            }
        }
    }
}
```

```
}

public function growText(pointSize:Number, maxLines:uint = 1):Number
{
    if (pointSize >= MAX_POINT_SIZE)
    {
        return pointSize;
    }

    this.changeSize(pointSize + 1);
    if (this.totalTextLines > maxLines)
    {
        // set it back to the last size
        this.changeSize(pointSize);
        return pointSize;
    }
    else
    {
        return growText(pointSize + 1, maxLines);
    }
}

public function shrinkText(pointSize:Number, maxLines:uint=1):Number
{
    if (pointSize <= MIN_POINT_SIZE)
    {
        return pointSize;
    }
    this.changeSize(pointSize);

    if (this.totalTextLines > maxLines)
    {
        return shrinkText(pointSize - 1, maxLines);
    }
    else
    {
        return pointSize;
    }
}
```

MultiColumnText.as 处理多列设计中的文本格式设置。它演示了 TextBlock 对象的灵活用法，该对象可作为用于创建文本行、设置文本行格式以及放置文本行的工厂。

```
package com.example.programmingas3.newslayout
{
    import flash.display.Sprite;
    import flash.text.engine.*;

    public class MultiColumnText extends Sprite
    {
        private var tb:TextBlock;
        private var te:TextElement;
        private var numColumns:uint = 2;
        private var gutter:uint = 10;
        private var leading:Number = 1.25;
        private var preferredWidth:Number = 400;
        private var preferredHeight:Number = 100;
        private var colWidth:int = 200;

        public function MultiColumnText(txt:String = "",cols:uint = 2,
            gutter:uint = 10, w:Number = 400, h:Number = 100,
            ef:ElementFormat = null):void
        {
            this.numColumns = Math.max(1, cols);
            this.gutter = Math.max(1, gutter);

            this.preferredWidth = w;
            this.preferredHeight = h;

            this.setColumnWidth();

            var field:FormattedTextBlock = new FormattedTextBlock(ef,txt,this.colWidth);
            var totLines:int = field.totalTextLines;
            field = null;
            var linesPerCol:int = Math.ceil(totLines/cols);

            tb = new TextBlock();
            te = new TextElement(txt,ef);
            tb.content = te;
            var textLine:TextLine = null;
            var x:Number = 0;
            var y:Number = 0;
            var i:int = 0;
            var j:int = 0;
            while (textLine = tb.createTextLine(textLine,this.colWidth,0,true))
            {
                textLine.x = Math.floor(i/(linesPerCol+1))*(this.colWidth+this.gutter);
                textLine.y = y;
                y += this.leading*textLine.height;
            }
        }
    }
}
```

```
j++;
if(j>linesPerCol)
{
    y = 0;
    j = 0;
}
i++;

this.addChild(textLine);
}

private function setColumnWidth():void
{
this.colWidth = Math.floor( (this.preferredWidth -
((this.numColumns - 1) * this.gutter)) / this.numColumns);
}

}
```

第 22 章：处理位图

除了矢量图功能以外，ActionScript 3.0 还提供创建位图图像或操作加载到 SWF 中的外部位图图像的像素数据的功能。使用访问和更改各个像素值的功能，您可以创建自己的滤镜式图像效果并使用内置杂点功能创建纹理和随机杂点。本章将介绍所有这些技术。

处理位图的基本知识

处理位图简介

使用数字图像时，您可能会遇到两种主要的图形类型：位图和矢量图形。位图图形也称为光栅图形，由排列为矩形网格形式的小方块（像素）组成。矢量图形由以数学方式生成的几何形状（如直线、曲线和多边形）组成。

位图图像用图像的宽度和高度来定义，以像素为量度单位，每个像素包含的位数表示像素包含的颜色数。在使用 RGB 颜色模型的位图图像中，像素由三个字节组成：红、绿和蓝。每个字节包含一个 0 至 255 之间的值。将字节与像素合并时，它们可以产生与艺术混合绘画颜色相似的颜色。例如，一个包含红色字节值 255、绿色字节值 102 和蓝色字节值 0 的像素可以形成明快的橙色。

位图图像的品质由图像分辨率和颜色深度位值共同确定。分辨率与图像中包含的像素数有关。像素数越大，分辨率越高，图像也就越精确。颜色深度与像素可包含的信息量有关。例如，颜色深度值为每像素 16 位的图像无法显示颜色深度为 48 位的图像所具有颜色数。因此，48 位图像与 16 位图像相比，其阴影具有更高的平滑度。

由于位图图形与分辨率有关，因此不能很好地进行缩放。当放大位图图像时，这一特性显得尤为突出。通常，放大位图有损其细节和品质。

位图文件格式

位图图像可分为几种常见的文件格式。这些格式使用不同类型的压缩算法减小文件大小，并基于图像的最终用途优化图像品质。Adobe Flash Player 和 Adobe AIR 支持的位图图像格式有 BMP、GIF、JPG、PNG 和 TIFF。

BMP

BMP（位映射）格式是 Microsoft Windows 操作系统使用的默认图像格式。这种格式不使用任何形式的压缩算法，因此文件大小通常较大。

GIF

图形交换格式（GIF）最初由 CompuServe 于 1987 年开发，作为一种传送 256 色（8 位颜色）图像的方式。此格式提供较小的文件大小，是基于 Web 的图像的理想格式。受此格式的调色板所限，GIF 图像通常不适用于照片，照片通常需要高度的阴影和颜色渐变。GIF 图像允许产生一位透明度，允许将颜色映射为清晰（或透明）。这可以使网页的背景颜色通过已映射透明度的图像显示出来。

JPEG

由联合图像专家组（JPEG）开发，JPEG（通常写成 JPG）图像格式使用有损压缩算法允许 24 位颜色深度具有很小的文件大小。有损压缩意味着每次保存图像，都会损失图像品质和数据，但会生成更小的文件大小。由于 JPEG 能够显示数百万计的颜色，因此它是照片的理想格式。控制应用于图像的压缩程度的功能使您能够控制图像品质和文件大小。

PNG

可移植网络图形 (PNG) 格式是作为受专利保护的 GIF 文件格式的开放源替代格式而开发的。PNG 最多支持 64 位颜色深度，允许使用最多 1600 万种颜色。由于 PNG 是一种比较新的格式，因此一些旧版本浏览器不支持 PNG 文件。与 JPG 不同，PNG 使用无损压缩，这意味着保存图像时不会丢失图像数据。PNG 文件还支持 Alpha 透明度，允许使用最多 256 级透明度。

TIFF

标签图像文件格式 (TIFF) 是在引入 PNG 之前的首选跨平台格式。TIFF 格式的缺点是，因为 TIFF 有多种不同变体，但没有一种阅读器能够处理所有版本。此外，所有 Web 浏览器当前均不支持这种格式。TIFF 可以使用有损或无损压缩，能够处理特定于设备的颜色空间（如 CMYK）。

透明位图和不透明位图

使用 GIF 或 PNG 格式的位图图像可以对每个像素添加一个额外字节（Alpha 通道）。此额外像素字节表示像素的透明度值。

GIF 图像允许使用一位透明度，这意味着您可以在 256 色调色板中指定一种透明的颜色。而 PNG 图像最多可以有 256 级透明度。当需要将图像或文本混合到背景中时，此功能特别有用。

ActionScript 3.0 在 `BitmapData` 类中复制了此额外透明度像素字节。与 PNG 透明度模型类似，`BitmapDataChannel.ALPHA` 常量最多提供 256 级透明度。

处理位图的常见任务

以下列表介绍了在 ActionScript 中处理位图图像时您要执行的几项任务：

- 在屏幕上显示位图
- 检索和设置像素颜色值
- 复制位图数据：
 - 创建与位图完全相同的副本
 - 从一个位图的一个颜色通道中向另一个位图的一个颜色通道中复制数据
 - 将屏幕快照显示对象复制到位图中
- 在位图图像中创建杂点和纹理
- 滚动位图

重要概念和术语

以下列表包含将在本章中遇到的重要术语：

- **Alpha**: 颜色或图像中的透明度级别（更准确地说是指不透明度）。Alpha 量通常称为“Alpha 通道”值。
- **ARGB 颜色**: 一种配色方案，其中每个像素的颜色都是红、绿和蓝色值的混合颜色，并且其透明度被指定为一个 Alpha 值。
- **颜色通道**: 通常将颜色表示为几种基本颜色的混合颜色；对于计算机图形来说，通常是红色、绿色和蓝色。每种基本颜色都视为一个颜色通道；每个颜色通道中的颜色量混合在一起可确定最终颜色。
- **颜色深度**: 也称为“位深度”，指专门用于每个像素的计算机内存量，因而可以确定图像中可以显示的可能颜色数。
- **像素**: 位图图像中的最小信息单位，实际上就是颜色点。
- **分辨率**: 图像的像素尺寸，它决定图像中包含的精细细节的级别。分辨率通常表示为用像素数表示的宽度和高度。
- **RGB 颜色**: 一种配色方案，其中每个像素的颜色均表示为红、绿和蓝色值的混合颜色。

完成本章中的示例

在学习本章的过程中，您可能想要测试示例代码。由于本章介绍的是创建和操作可视内容，因此测试代码涉及到运行代码以及在创建的 SWF 中查看结果。

要测试本章中的代码示例，请执行以下操作：

- 1 使用 Flash 创作工具创建空文档
- 2 在时间轴上选择一关键帧。
- 3 打开“动作”面板，将代码复制到“脚本”窗格中。
- 4 通过“控制”>“测试影片”运行程序。

您将在所创建的 SWF 文件中看到代码的结果。

几乎所有示例代码都包括用于创建位图图像的代码，因此您可以直接测试代码，而无需提供任何位图内容。或者，如果您希望在自己的图像上测试代码，则可以将该图像导入 Adobe Flash CS4 Professional，或将外部图像加载到测试 SWF 并将其位图数据与示例代码一起使用。有关加载外部图像的说明，请参阅第 257 页的“[动态加载显示内容](#)”。

Bitmap 和 BitmapData 类

使用位图图像的主要 ActionScript 3.0 类有 [Bitmap](#) 类和 [BitmapData](#) 类；前者用于在屏幕上显示位图图像，后者用于访问和操作位图的原始图像数据。

了解 Bitmap 类

作为 [DisplayObject](#) 类的子类，[Bitmap](#) 类是用于显示位图图像的主要 ActionScript 3.0 类。这些图像可能已经通过 [flash.display.Loader](#) 类加载到 Flash Player 或 Adobe AIR 中，或已经使用 [Bitmap\(\)](#) 构造函数动态创建。从外部源加载图像时，[Bitmap](#) 对象只能使用 GIF、JPEG 或 PNG 格式的图像。实例化后，可将 [Bitmap](#) 实例视为需要呈现在舞台上的 [BitmapData](#) 对象的包装。由于 [Bitmap](#) 实例是一个显示对象，因此可以使用显示对象的所有特性和功能来操作 [Bitmap](#) 实例。有关使用显示对象的详细信息，请参阅第 224 页的“[显示编程](#)”。

像素贴紧和平滑

除了所有显示对象常见的功能外，[Bitmap](#) 类还提供了特定于位图图像的一些附加功能。

与 Flash 创作工具中的贴紧像素功能类似，[Bitmap](#) 类的 [pixelSnapping](#) 属性可确定 [Bitmap](#) 对象是否贴紧最近的像素。此属性接受 [PixelSnapping](#) 类中定义的三个常量之一：ALWAYS、AUTO 和 NEVER。

应用像素贴紧的语法为：

```
myBitmap.pixelSnapping = PixelSnapping.ALWAYS;
```

通常，缩放位图图像时，图像会变得模糊或扭曲。若要帮助减少这种扭曲，请使用 [BitmapData](#) 类的 [smoothing](#) 属性。如果将该布尔值属性设置为 `true`，当缩放图像时，可使图像中的像素平滑或消除锯齿。它可使图像更加清晰、更加自然。

了解 BitmapData 类

[BitmapData](#) 类位于 [flash.display](#) 包中，它可以看作是加载的或动态创建的位图图像中包含的像素的照片快照。此快照用对象中的像素数据的数组表示。[BitmapData](#) 类还包含一系列内置方法，可用于创建和处理像素数据。

若要实例化 [BitmapData](#) 对象，请使用以下代码：

```
var myBitmap:BitmapData = new BitmapData(width:Number, height:Number, transparent:Boolean,  
fillColor:uint);
```

`width` 和 `height` 参数指定位图的大小；二者的最大值都是 2880 像素。`transparent` 参数指定位图数据是 (true) 否 (false) 包括 Alpha 通道。`fillColor` 参数是一个 32 位颜色值，它指定背景颜色和透明度值（如果设置为 true）。以下示例创建一个具有 50% 透明的橙色背景的 `BitmapData` 对象：

```
var myBitmap:BitmapData = new BitmapData(150, 150, true, 0x80FF3300);
```

若要在屏幕上呈现新创建的 `BitmapData` 对象，请将此对象分配给或包装到 `Bitmap` 实例中。为此，可以将 `BitmapData` 对象作为 `Bitmap` 对象的构造函数参数传递，也可以将此对象分配给现有 `Bitmap` 实例的 `bitmapData` 属性。您还必须通过调用将包含该 `Bitmap` 实例的显示对象容器的 `addChild()` 或 `addChildAt()` 方法将该 `Bitmap` 实例添加到显示列表中。有关使用显示列表的详细信息，请参阅第 231 页的“[在显示列表中添加显示对象](#)”。

以下示例创建一个具有红色填充的 `BitmapData` 对象，并在 `Bitmap` 实例中显示此对象：

```
var myBitmapDataObject:BitmapData = new BitmapData(150, 150, false, 0xFF0000);
var myImage:Bitmap = new Bitmap(myBitmapDataObject);
addChild(myImage);
```

处理像素

`BitmapData` 类包含一组用于处理像素数据值的方法。

处理单个像素

在像素级别更改位图图像的外观时，您首先需要获取要处理的区域中包含的像素的颜色值。使用 `getPixel()` 方法可读取这些像素值。

`getPixel()` 方法从作为参数传递的一组 `x, y`（像素）坐标中检索 RGB 值。如果您要处理的像素包括透明度（Alpha 通道）信息，则需要使用 `getPixel32()` 方法。此方法也可以检索 RGB 值，但与 `getPixel()` 不同，`getPixel32()` 返回的值包含表示所选像素的 Alpha 通道（透明度）值的附加数据。

或者，如果只想更改位图中包含的某个像素的颜色或透明度，则可以使用 `setPixel()` 或 `setPixel32()` 方法。若要设置像素的颜色，只需将 `x`、`y` 坐标和颜色值传递给这两种方法之一即可。

以下示例使用 `setPixel()` 在绿色 `BitmapData` 背景上绘制交叉形状。然后，此示例使用 `getPixel()` 从坐标 50, 50 处的像素中检索颜色值并跟踪返回的值。

```
import flash.display.Bitmap;
import flash.display.BitmapData;

var myBitmapData:BitmapData = new BitmapData(100, 100, false, 0x009900);

for (var i:uint = 0; i < 100; i++)
{
    var red:uint = 0xFF0000;
    myBitmapData.setPixel(50, i, red);
    myBitmapData.setPixel(i, 50, red);
}

var myBitmapImage:Bitmap = new Bitmap(myBitmapData);
addChild(myBitmapImage);

var pixelValue:uint = myBitmapData.getPixel(50, 50);
trace(pixelValue.toString(16));
```

如果要读取一组像素而不是单个像素的值，请使用 `getPixels()` 方法。此方法从作为参数传递的矩形像素数据区域中生成字节数组。字节数组的每个元素（即像素值）都是无符号的整数（32 位未经相乘的像素值）。

相反，若要更改（或设置）一组像素的值，请使用 `setPixels()` 方法。此方法需要联合使用两个参数（`rect` 和 `inputByteArray`）来输出像素数据（`inputByteArray`）的矩形区域（`rect`）。

从 `inputByteArray` 中读取（或写入）数据时，会为数组中的每个像素调用 `ByteArray.readUnsignedInt()` 方法。如果由于某些原因，`inputByteArray` 未包含像素数据的整个矩形，则该方法会停止处理该点处的图像数据。

必须记住的是，对于获取和设置像素数据，字节数组需要有 32 位 Alpha、红、绿、蓝（ARGB）像素值。

以下示例使用 `getPixels()` 和 `setPixels()` 方法将一组像素从一个 `BitmapData` 对象复制到另一个对象：

```
import flash.display.Bitmap;
import flash.display.BitmapData;
import flash.utils.ByteArray;
import flash.geom.Rectangle;

var bitmapDataObject1:BitmapData = new BitmapData(100, 100, false, 0x006666FF);
var bitmapDataObject2:BitmapData = new BitmapData(100, 100, false, 0x00FF0000);

var rect:Rectangle = new Rectangle(0, 0, 100, 100);
var bytes:ByteArray = bitmapDataObject1.getPixels(rect);

bytes.position = 0;
bitmapDataObject2.setPixels(rect, bytes);

var bitmapImage1:Bitmap = new Bitmap(bitmapDataObject1);
addChild(bitmapImage1);
var bitmapImage2:Bitmap = new Bitmap(bitmapDataObject2);
addChild(bitmapImage2);
bitmapImage2.x = 110;
```

像素级别冲突检测

`BitmapData.hitTest()` 方法可以在位图数据和另一个对象或点之间执行像素级别冲突检测。

`BitmapData.hitTest()` 方法接受五个参数：

- `firstPoint (Point)`: 此参数指在其上执行点击测试的第一个 `BitmapData` 的左上角的像素位置。
- `firstAlphaThreshold (uint)`: 此参数指定对于此点击测试视为不透明的最高 Alpha 通道值。
- `secondObject (Object)`: 此参数表示影响区域。`secondObject` 对象可以是 `Rectangle`、`Point`、`Bitmap` 或 `BitmapData` 对象。此对象表示在其上执行冲突检测的点击区域。
- `secondBitmapDataPoint (Point)`: 此可选参数用于在第二个 `BitmapData` 对象中定义像素位置。只有当 `secondObject` 的值为 `BitmapData` 对象时，才使用此参数。默认值为 `null`。
- `secondAlphaThreshold (uint)`: 此可选参数表示在第二个 `BitmapData` 对象中视为不透明的最高 Alpha 通道值。默认值为 1。仅当 `secondObject` 的值是 `BitmapData` 对象并且这两个 `BitmapData` 对象都透明时，才使用此参数。

在不透明图像上执行冲突检测时，请牢记，ActionScript 会将图像视为完全不透明的矩形（或边框）。或者，在透明的图像上执行像素级别点击测试时，需要两个图像都是透明的。除此之外，ActionScript 还使用 Alpha 阈值参数来确定像素在哪点开始从透明变为不透明。

以下示例创建三个位图图像并使用两个不同冲突点（一个返回 `false`，另一个返回 `true`）检查像素冲突：

```

import flash.display.Bitmap;
import flash.display.BitmapData;
import flash.geom.Point;

var bmd1:BitmapData = new BitmapData(100, 100, false, 0x000000FF);
var bmd2:BitmapData = new BitmapData(20, 20, false, 0x00FF3300);

var bm1:Bitmap = new Bitmap(bmd1);
this.addChild(bm1);

// Create a red square.
var redSquare1:Bitmap = new Bitmap(bmd2);
this.addChild(redSquare1);
redSquare1.x = 0;

// Create a second red square.
var redSquare2:Bitmap = new Bitmap(bmd2);
this.addChild(redSquare2);
redSquare2.x = 150;
redSquare2.y = 150;

// Define the point at the top-left corner of the bitmap.
var pt1:Point = new Point(0, 0);
// Define the point at the center of redSquare1.
var pt2:Point = new Point(20, 20);
// Define the point at the center of redSquare2.
var pt3:Point = new Point(160, 160);

trace(bmd1.hitTest(pt1, 0xFF, pt2)); // true
trace(bmd1.hitTest(pt1, 0xFF, pt3)); // false

```

复制位图数据

若要从一个图像向另一个图像中复制位图数据，可以使用多种方法：`clone()`、`copyPixels()`、`copyChannel()` 和 `draw()`。

正如名称的含义一样，`clone()` 方法允许您将位图数据从一个 `BitmapData` 对象克隆或采样到另一个对象。调用此方法时，此方法返回一个新的 `BitmapData` 对象，它是与被复制的原始实例完全一样的克隆。

以下示例克隆橙色（父级）正方形的一个副本，并将克隆放在原始父级正方形的旁边：

```

import flash.display.Bitmap;
import flash.display.BitmapData;

var myParentSquareBitmap:BitmapData = new BitmapData(100, 100, false, 0x00ff3300);
var myClonedChild:BitmapData = myParentSquareBitmap.clone();

var myParentSquareContainer:Bitmap = new Bitmap(myParentSquareBitmap);
this.addChild(myParentSquareContainer);

var myClonedChildContainer:Bitmap = new Bitmap(myClonedChild);
this.addChild(myClonedChildContainer);
myClonedChildContainer.x = 110;

```

`copyPixels()` 方法是一种将像素从一个 `BitmapData` 对象复制到另一个 `BitmapData` 对象的快速且简便的方法。该方法会拍摄源图像的矩形快照（由 `sourceRect` 参数定义），并将其复制到另一个矩形区域（大小相等）。新“粘贴”的矩形的位置在 `destPoint` 参数中定义。

`copyChannel()` 方法从源 `BitmapData` 对象中采集预定义的颜色通道值（Alpha、红、绿或蓝），并将此值复制到目标 `BitmapData` 对象的通道中。调用此方法不会影响目标 `BitmapData` 对象中的其它通道。

`draw()` 方法将源 `sprite`、影片剪辑或其它显示对象中的图形内容绘制或呈现在新位图上。使用 `matrix`、`colorTransform`、`blendMode` 和目标 `clipRect` 参数，可以修改新位图的呈现方式。此方法使用 Flash Player 和 AIR 中的矢量渲染器生成数据。

调用 `draw()` 时，需要将源对象（`sprite`、影片剪辑或其它显示对象）作为第一个参数传递，如下所示：

```
myBitmap.draw(movieClip);
```

如果源对象在最初加载后应用了变形（颜色、矩阵等等），则不能将这些变形复制到新对象。如果想要将变形复制到新位图，则需要将 `transform` 属性的值从原始对象复制到使用新 `BitmapData` 对象的 `Bitmap` 对象的 `transform` 属性中。

使用杂点功能制作纹理

若要修改位图的外观，可以使用 `noise()` 方法或 `perlinNoise()` 方法对位图应用杂点效果。可以把杂点效果比作未调谐的电视屏幕的静态外观。

若要对位图应用杂点效果，请使用 `noise()` 方法。此方法对位图图像的指定区域中的像素应用随机颜色值。

此方法接受五个参数：

- `randomSeed (int)`: 决定图案的随机种子数。不管名称具有什么样的含义，只要传递的数字相同，此数字就会生成相同的结果。为了获得真正的随机结果，请使用 `Math.random()` 方法为此参数传递随机数字。
- `low (uint)`: 此参数指要为每个像素生成的最低值（0 至 255）。默认值为 0。将此参数设置为较低值会产生较暗的杂点图案，而将此参数设置为较高值会产生较亮的图案。
- `high (uint)`: 此参数指要为每个像素生成的最高值（0 至 255）。默认值为 255。将此参数设置为较低值会产生较暗的杂点图案，而将此参数设置为较高值会产生较亮的图案。
- `channelOptions (uint)`: 此参数指定将向位图对象的哪个颜色通道应用杂点图案。此数字可以是四个颜色通道 ARGB 值的任意组合。默认值为 7。
- `grayScale (Boolean)`: 设置为 `true` 时，此参数对位图像素应用 `randomSeed` 值，可有效地褪去图像中的所有颜色。此参数不影响 Alpha 通道。默认值是 `false`。

以下示例创建一个位图图像，并对它应用蓝色杂点图案：

```
import flash.display.Bitmap;
import flash.display.BitmapData;

var myBitmap:BitmapData = new BitmapData(250, 250, false, 0xff000000);
myBitmap.noise(500, 0, 255, BitmapDataChannel.BLUE, false);
var image:Bitmap = new Bitmap(myBitmap);
addChild(image);
```

如果要创建更好的有机外观纹理，请使用 `perlinNoise()` 方法。`perlinNoise()` 方法可生成逼真、有机的纹理，是用于烟雾、云彩、水、火或爆炸的理想图案。

由于 `perlinNoise()` 方法是由算法生成的，因此它使用的内存比基于位图的纹理少。但还是会处理器的使用有影响，特别是对于旧计算机，会降低 Flash 所创建的内容的处理速度，使屏幕重绘的速度比帧速率慢。这主要是因为需要进行浮点计算，以便处理 Perlin 杂点算法。

此方法接受九个参数（前六个是必需参数）：

- `baseX (Number)`: 决定创建的图案的 x（大小）值。
- `baseY (Number)`: 决定创建的图案的 y（大小）值。
- `numOctaves (uint)`: 要组合以创建此杂点的 `octave` 函数或各个杂点函数的数目。`octave` 数目越大，创建的图像越精细，但这需要更多的处理时间。

- **randomSeed (int):** 随机种子数的功能与在 noise() 函数中的功能完全相同。为了获得真正的随机结果，请使用 Math.random() 方法为此参数传递随机数字。
- **stitch (Boolean):** 如果设置为 true，则此方法尝试缝合（或平滑）图像的过渡边缘以形成无缝的纹理，用于作为位图填充进行平铺。
- **fractalNoise (Boolean):** 此参数与此方法生成的渐变的边缘有关。如果设置为 true，则此方法生成的碎片杂点会对效果的边缘进行平滑处理。如果设置为 false，则将生成湍流。带有湍流的图像具有可见的不连续性渐变，可以使用它处理更接近锐化的视觉效果，例如，火焰或海浪。
- **channelOptions (uint):** channelOptions 参数的功能与在 noise() 方法中的功能完全相同。它指定对哪个颜色通道（在位图上）应用杂点图案。此数字可以是四个颜色通道 ARGB 值的任意组合。默认值为 7。
- **grayScale (Boolean):** grayScale 参数的功能与在 noise() 方法中的功能完全相同。如果设置为 true，则对位图像素应用 randomSeed 值，可有效地褪去图像中的所有颜色。默认值是 false。
- **offsets (Array):** 对应于每个 octave 的 x 和 y 偏移的点数组。通过处理偏移值，可以平滑滚动图像层。偏移数组中的每个点将影响一个特定的 octave 杂点函数。默认值为 null。

以下示例创建一个 150 x 150 像素的 BitmapData 对象，该对象调用 perlinNoise() 方法来生成绿色和蓝色的云彩效果：

```
import flash.display.Bitmap;
import flash.display.BitmapData;

var myBitmapDataObject:BitmapData = new BitmapData(150, 150, false, 0x00FF0000);

var seed:Number = Math.floor(Math.random() * 100);
var channels:uint = BitmapDataChannel.GREEN | BitmapDataChannel.BLUE
myBitmapDataObject.perlinNoise(100, 80, 6, seed, false, true, channels, false, null);

var myBitmap:Bitmap = new Bitmap(myBitmapDataObject);
addChild(myBitmap);
```

滚动位图

设想您创建了一个街道图应用程序，每次用户移动该图时，都需要您更新视图（即使该图只移动了几个像素）。

创建此功能的一种方式是，每次用户移动街道图时，均重新呈现包含更新的街道图视图的新图像。或者，创建一个大型图像，并使用 scroll() 方法。

scroll() 方法可以复制屏幕上的位图，然后将它粘贴到由 (x, y) 参数指定的新偏移位置。如果位图的一部分恰巧在舞台以外，则会产生图像发生移位的效果。与计时器函数（或 enterFrame 事件）配合使用时，可以使图像呈现动画或滚动效果。

以下示例采用前面的 Perlin 杂点示例并生成较大的位图图像（其四分之三呈现在舞台外面）。然后应用 scroll() 方法和一个 enterFrame 事件侦听器，使图像在对角线向下方向偏移一个像素。每次输入帧时均会调用此方法，因此，随着图像向下滚动，图像位于屏幕以外的部分会呈现在舞台上。

```
import flash.display.Bitmap;
import flash.display.BitmapData;

var myBitmapDataObject:BitmapData = new BitmapData(1000, 1000, false, 0x00FF0000);
var seed:Number = Math.floor(Math.random() * 100);
var channels:uint = BitmapDataChannel.GREEN | BitmapDataChannel.BLUE;
myBitmapDataObject.perlinNoise(100, 80, 6, seed, false, true, channels, false, null);

var myBitmap:Bitmap = new Bitmap(myBitmapDataObject);
myBitmap.x = -750;
myBitmap.y = -750;
addChild(myBitmap);

addEventListener(Event.ENTER_FRAME, scrollBitmap);

function scrollBitmap(event:Event):void
{
    myBitmapDataObject.scroll(1, 1);
}
```

利用 mipmap 处理

“MIP map”（也称为“mipmap”）是组合在一起并与纹理关联的位图，可改善运行时呈现品质和性能。Flash Player 9.0.115.0 和更高版本及 AIR 通过创建每个位图的不同缩放（从 50% 开始）的优化版本，来实现此技术（该过程称为“mipmap 处理”）。

Flash Player 和 AIR 为使用 ActionScript 3.0 [Loader](#) 类显示的位图（JPEG、GIF 或 PNG 文件）、Flash 创作工具库中的位图或 [BitmapData](#) 对象创建 MIP map。Flash Player 为使用 ActionScript 2.0 [loadMovie\(\)](#) 函数显示的位图创建 MIP map。

MIP map 不适用于应用滤镜的对象或缓存位图的影片剪辑。不过，如果应用滤镜的显示对象中包含位图转换，则即使位图位于被遮罩的内容中，也会应用 MIP map。

Flash Player 和 AIR 的 mipmap 处理是自动执行的，但您可以遵循几条准则，以确保您的图像利用此优化技术：

- 对于视频播放，请将 Video 对象的 [smoothing](#) 属性设置为 true（请参阅 [Video](#) 类）。
- 对于位图，不一定要将 [smoothing](#) 属性设置为 true，但当位图使用平滑处理时品质的改善更为显著。
- 对于二维图像，请使用可被 4 或 8 整除的位图大小（如 640 x 128，可按如下方式递减：320 x 64 > 160 x 32 > 80 x 16 > 40 x 8 > 20 x 4 > 10 x 2 > 5 x 1）；对于三维纹理，请使用 2^n 的位图大小。生成 MIP map 的位图的宽和高均为 2^n （如 256 x 256、512 x 512、1024 x 1024）。当 Flash Player 或 AIR 遇到宽或高为奇数的位图时，会停止 Mipmap 处理。

示例：旋转的月球动画

旋转的月球动画示例说明了处理位图对象和位图图像数据（[BitmapData](#) 对象）的方法。该示例使用月球表面的平面图像作为原始图像数据来创建一个旋转的月球动画。将对以下方法进行说明：

- 加载外部图像并访问其原始图像数据
- 通过重复复制源图像不同部分的像素创建动画
- 通过设置像素值创建位图图像

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可以在 Samples/SpinningMoon 文件夹中找到“旋转的月球动画”的应用程序文件。该应用程序包含以下文件：

文件	描述
SpinningMoon.mxml 或 SpinningMoon.fla	Flex (MXML) 或 Flash (FLA) 中的主应用程序文件。
com/example/programmingas3/moon/MoonSphere.as	用于执行加载、显示月球和创建月球动画的功能的类。
moonMap.png	包含月球表面照片的图像文件（将加载此图像文件并使用它来创建旋转的月球动画）。

将外部图像作为位图数据加载

此范例执行的第一项主要任务是加载外部图像文件，即月球表面照片。加载操作由 **MoonSphere** 类中的两个方法处理：**MoonSphere()** 构造函数（启动加载过程）和 **imageLoadComplete()** 方法（完成外部图像加载后调用该方法）。

加载外部图像与加载外部 SWF 类似，两者都使用 **flash.display.Loader** 类的实例执行加载操作。启动图像加载的 **MoonSphere()** 方法中的实际代码如下：

```
var imageLoader:Loader = new Loader();
imageLoader.contentLoaderInfo.addEventListener(Event.COMPLETE, imageLoadComplete);
imageLoader.load(new URLRequest("moonMap.png"));
```

第一行声明名为 **imageLoader** 的 **Loader** 实例。第三行通过调用 **Loader** 对象的 **load()** 方法并传递一个 **URLRequest** 实例（表示要加载的图像的 URL）来实际启动加载过程。第二行设置在完成图像加载时将触发的事件侦听器。请注意：不是对 **Loader** 实例本身调用 **addEventListener()** 方法；而是对 **Loader** 对象的 **contentLoaderInfo** 属性调用该方法。**Loader** 实例本身不会调度与所加载内容相关的事件。然而，它的 **contentLoaderInfo** 属性包含对 **LoaderInfo** 对象的引用，该对象与加载到 **Loader** 对象的内容（本例中为外部图像）相关联。该 **LoaderInfo** 对象提供与外部内容加载进度及加载完成有关的几个事件，其中包括 **complete** 事件（**Event.COMPLETE**），该事件将在完成图像加载时触发对 **imageLoadComplete()** 方法的调用。


启动外部图像加载是加载过程中的重要部分，而了解加载完成时要执行什么操作也同等重要。如以上代码所示，完成加载图像后将调用 **imageLoadComplete()** 函数。该函数对加载的图像数据执行一些操作，这将在后续部分中进行说明。然而，若要使用图像数据，还需要访问该数据。当使用某 **Loader** 对象来加载外部图像时，加载的图像将变为一个 **Bitmap** 实例，并附加为该 **Loader** 对象的一个子显示对象。在本例中，**Loader** 实例可作为事件对象的一部分用于事件侦听器方法，此事件对象将作为参数传递给该方法。**imageLoadComplete()** 方法的前几行如下：

```
private function imageLoadComplete(event:Event):void
{
    textureMap = event.target.content.bitmapData;
    ...
}
```

请注意，事件对象参数名为 **event**，它是 **Event** 类的一个实例。**Event** 类的每个实例都具有一个 **target** 属性，该属性将引用触发事件的对象（本例中为 **LoaderInfo** 实例，如前所述，将对该实例调用 **addEventListener()** 方法。）而 **LoaderInfo** 对象又具有一个 **content** 属性，加载过程完成后，该属性将包含 **Bitmap** 实例，其中具有加载的位图图像。如果要在屏幕上直接显示该图像，则可以将此 **Bitmap** 实例 (**event.target.content**) 附加到一个显示对象容器。（也可以将 **Loader** 对象附加到一个显示对象容器。）但是，在该示例中，加载的内容将用作原始图像数据的源而不是显示在屏幕上。因此，**imageLoadComplete()** 方法的第一行读取加载的 **Bitmap** 实例的 **bitmapData** 属性 (**event.target.content.bitmapData**)，并将其存储在名为 **textureMap** 的实例变量（在以下部分中介绍）中，该变量用作创建旋转的月球动画的图像数据源。

通过复制像素创建动画

动画的基本定义是通过随时间变化改变图像而产生的运动或变化的视觉效果。此示例的目标是创建月球绕其垂直轴旋转的视觉效果。然而，对于动画而言，您可以忽略该示例中球形扭曲方面的问题。假设已加载并用作月球图像数据源的实际图像如下：



如您所见，该图像并不是一个或几个球体；它是月球表面的一张矩形照片。因为该照片刚好是在月球赤道上拍摄的，所以图像中靠近图像顶部和底部的部分发生拉伸和扭曲。若要消除图像的扭曲使其具有球形外观，我们将使用置换图滤镜（在后面进行介绍）。但是，因为该源图像是矩形的，所以若要产生旋转球体的视觉效果，代码只要能完成水平滑过月球表面照片的操作即可（在下面几段中进行介绍）。

请注意，该图像实际上由月球表面照片的两个副本彼此相接而成。该图像是要从中重复复制图像数据来创建动画外观的源图像。通过两个图像副本彼此相接，会更容易产生连续、不间断的滚动效果。让我们逐步浏览动画生成的过程来看看这是如何实现的。

该过程实际上涉及两个单独的 ActionScript 对象。首先，有加载的源图像，在代码中由名为 textureMap 的 BitmapData 实例表示。如前所述，外部图像加载后，将用图像数据来填充 textureMap，使用的代码如下：

```
textureMap = event.target.content.bitmapData;
```


textureMap 的内容就是前面显示的图像。另外，为了产生旋转动画，该示例使用名为 sphere 的 Bitmap 实例，该实例是在屏幕上显示月球图像的实际显示对象。与 textureMap 一样，sphere 对象也是在 imageLoadComplete() 方法中创建并使用其初始图像数据填充，使用的代码如下：

```
sphere = new Bitmap();
sphere.bitmapData = new BitmapData(textureMap.width / 2, textureMap.height);
sphere.bitmapData.copyPixels(textureMap,
    new Rectangle(0, 0, sphere.width, sphere.height),
    new Point(0, 0));
```

如代码所示，sphere 被实例化了。其 bitmapData 属性（通过 sphere 显示的原始图像数据）具有与 textureMap 相同的高度和一半的宽度。换句话说，sphere 的内容将是一幅月球照片的大小（因为 textureMap 图像包含并排的两幅月球照片）。接下来，用图像数据填充 bitmapData 属性，填充时使用的是 copyPixels() 方法。copyPixels() 方法调用中的参数指明以下几点：

- 第一个参数指明从 textureMap 复制图像数据。
- 第二个参数（新的 Rectangle 实例）指定图像快照应该从 textureMap 的哪部分拍摄；在本例中，快照是从 textureMap 左上角开始的一个矩形（由前两个 Rectangle() 参数 0,0 指明），矩形快照的宽度和高度与 sphere 的 width 和 height 属性一致。
- 第三个参数（新的 Point 实例）的 x 和 y 值都为 0，它定义了像素数据的目标位置——本例中为 sphere.bitmapData 的左上角 (0, 0)。

从视觉表示形式上看，该代码将复制下图中用轮廓线标出的、`textureMap` 的像素，并将其粘贴到 `sphere` 上。换句话说，`sphere` 的 `BitmapData` 内容是这里突出显示的 `textureMap` 部分：



然而请记住，这只是 `sphere` 的初始状态 — 复制到 `sphere` 上的第一项图像内容。

加载源图像并创建 `sphere` 之后，由 `imageLoadComplete()` 方法执行的最终任务是设置动画。动画由名为 `rotationTimer` 的 `Timer` 实例驱动，该实例由以下代码创建并启动：

```
var rotationTimer:Timer = new Timer(15);
rotationTimer.addEventListener(TimerEvent.TIMER, rotateMoon);
rotationTimer.start();
```

代码首先创建名为 `rotationTimer` 的 `Timer` 实例；传递给 `Timer()` 构造函数的参数指明 `rotationTimer` 应每 15 毫秒触发一次其 `timer` 事件。接下来将调用 `addEventListener()` 方法，以指定在发生 `timer` 事件 (`TimerEvent.TIMER`) 时调用 `rotateMoon()` 方法。最后，计时器实际上是通过调用其 `start()` 方法启动的。

根据 `rotationTimer` 的定义方式，约每 15 毫秒 Flash Player 调用一次 `MoonSphere` 类中的 `rotateMoon()` 方法（用于产生月球动画）。`rotateMoon()` 方法的源代码如下：

```
private function rotateMoon(event:TimerEvent):void
{
    sourceX += 1;
    if (sourceX > textureMap.width / 2)
    {
        sourceX = 0;
    }

    sphere.bitmapData.copyPixels(textureMap,
        new Rectangle(sourceX, 0, sphere.width, sphere.height),
        new Point(0, 0));


    event.updateAfterEvent();
}
```

该代码实现以下三方面的操作：

- 1 变量 `sourceX` 的值（最初设为 0）增加 1。

```
sourceX += 1;
```

您将看到，`sourceX` 用于确定 `textureMap` 中的位置（从该位置将像素复制到 `sphere`），因此该代码会产生在 `textureMap` 上将矩形向右移动一个像素的效果。返回到视觉表示形式，经过几个动画循环之后，源矩形将向右移动几个像素，如下所示：




经过几个循环之后，矩形将进一步移动：



像素复制位置的这种平稳渐进式移动是动画制作的关键。通过缓慢、连续地将源位置移动到右侧，`sphere` 中显示在屏幕上的图像显示为连续地滑向左侧。这就是源图像 (`textureMap`) 需要两个月球表面照片副本的原因。由于矩形连续移动到右侧，因此大多数时间该矩形不是在一张月球照片上而是与两张月球照片发生重叠。


- 2 随着源矩形缓慢移到右侧，会出现一个问题。最后，矩形将到达 `textureMap` 的右边缘，它将用完要复制到 `sphere` 上的月球照片像素：



下一行代码将解决这个问题：

```
if (sourceX >= textureMap.width / 2)
{
    sourceX = 0;
}
```

该代码检查 `sourceX` (矩形的左边缘) 是否已到达 `textureMap` 的中部。如果是，它会将 `sourceX` 重置为 0，将其移回到 `textureMap` 的左边缘并重新开始循环：



- 3** 计算出适当的 `sourceX` 值后，创建动画的最后一步是将新的源矩形像素实际复制到 `sphere` 上。实现这一操作的代码与最初填充 `sphere` 的代码（如前所述）非常类似；唯一的不同是：在本例的 `new Rectangle()` 构造函数调用中，矩形的左边缘位于 `sourceX`：

```
sphere.bitmapData.copyPixels(textureMap,
    new Rectangle(sourceX, 0, sphere.width, sphere.height),
    new Point(0, 0));
```

请记住，此代码每 15 毫秒重复调用一次。由于源矩形的位置是连续移动的，并且像素被复制到 `sphere` 上，因此在屏幕上显示为由 `sphere` 表示的月球照片图像发生连续滑动。换句话说，月球显示为连续旋转。

创建球形外观

当然，月球是一个球体而不是一个矩形。由于要形成连续动画，因此该示例需要拍摄矩形的月球表面照片，并将其转换成球体。这涉及两个单独的步骤：一个用于隐藏月球表面照片中除圆形区域之外的所有内容的遮罩，以及一个用于扭曲月球照片外观使其具有三维外观的置换图滤镜。

首先，圆形遮罩用于隐藏 `MoonSphere` 对象中除通过滤镜创建的球体之外的所有内容。以下代码创建一个作为 `Shape` 实例的遮罩，并将其应用为 `MoonSphere` 实例的遮罩：

```
moonMask = new Shape();
moonMask.graphics.beginFill(0);
moonMask.graphics.drawCircle(0, 0, radius);
this.addChild(moonMask);
this.mask = moonMask;
```

请注意，由于 `MoonSphere` 是一个显示对象（它基于 `Sprite` 类），因此可以使用该显示对象继承的 `mask` 属性将遮罩直接应用于 `MoonSphere` 实例。



仅使用圆形遮罩隐藏照片的某些部分不足以创建逼真的旋转球体效果。由于月球表面照片拍摄方式的限制，导致照片的尺寸不成比例；与赤道上的部分相比，图像上越靠近图像顶部或底部的部分扭曲和拉伸得越严重。为了对月球照片的外观进行变形以使其具有三维效果，我们将使用置换图滤镜。

置换图滤镜是一种用于扭曲图像的滤镜。在本例中，将通过水平挤压图像的顶部和底部而保持中部不变来“扭曲”月球照片，从而使其看起来更加逼真。假设对照片的正方形部分执行滤镜操作，挤压顶部和底部而不挤压中部将使正方形变为圆形。为该扭曲图像添加动画效果所产生的另一效果是：与靠近顶部和底部的区域相比，图像的中部看起来所移动的实际像素距离更大，这将产生圆实际上是一个三维对象（球体）的视觉效果。

以下代码用于创建名为 `displaceFilter` 的置换图滤镜：

```
var displaceFilter:DisplacementMapFilter;
displaceFilter = new DisplacementMapFilter(fisheyeLens,
                                             new Point(radius, 0),
                                             BitmapDataChannel.RED,
                                             BitmapDataChannel.GREEN,
                                             radius, 0);
```

第一个参数 `fisheyeLens` 称为置换图图像；在本例中，它是以编程方式创建的 `BitmapData` 对象。创建该图像的说明见后面的第 419 页的“[通过设置像素值创建位图图像](#)”一节。其它参数说明过滤后的图像中应该应用滤镜的位置、使用哪些颜色通道来控制置换效果以及将在何种范围内对置换产生影响。一旦创建置换图滤镜，它将应用于仍位于 `imageLoadComplete()` 方法中的 `sphere`：

```
sphere.filters = [displaceFilter];
```

应用了遮罩和置换图滤镜的最终图像如下所示：




每个旋转月球动画循环之后，`sphere` 的 `BitmapData` 内容将由新的源图像数据快照覆盖。但是，无需每次都重新应用滤镜。这是因为滤镜应用到了 `Bitmap` 实例（显示对象）而不是位图数据（原始像素信息）。请记住，`Bitmap` 实例不是实际的位图数据；它是在屏幕上显示位图数据的显示对象。举例来说，`Bitmap` 实例就像用于在屏幕上显示照片幻灯片的幻灯片放映机，而 `BitmapData` 对象就像可以通过幻灯片放映机显示的实际照片幻灯片。滤镜可以直接应用于 `BitmapData` 对象，这与直接在照片幻灯片上绘图以更改图像相似。滤镜也可以应用于任何显示对象（包括 `Bitmap` 实例）；这与在幻灯片放映机的镜头前面放一个滤镜以使屏幕上显示的输出变形（丝毫不改变原始幻灯片）相似。因为可通过 `Bitmap` 实例的 `bitmapData` 属性来访问原始位图数据，所以滤镜可能已直接应用于原始位图数据。然而，在本例中，将滤镜应用于 `Bitmap` 显示对象比应用于位图数据更有意义。

有关在 ActionScript 中使用置换图滤镜的详细信息，请参阅第 293 页的“[过滤显示对象](#)”。

通过设置像素值创建位图图像

置换图滤镜的一个重要方面是它实际上涉及两个图像。一个图像为源图像，即由滤镜实际更改的图像。在该示例中，源图像是名为 `sphere` 的 `Bitmap` 实例。滤镜所用的另一个图像称为映射图像。映射图像不实际显示在屏幕上。相反，它的每个像素的颜色都用作置换函数的输入 — 置换图图像中位于特定 `x`、`y` 坐标的像素的颜色决定应用于源图像中位于该 `x`、`y` 坐标的像素的置换量（物理位移）。

因此，为了使用置换图滤镜创建球体效果，该范例需要合适的置换图图像 — 一个包含灰色背景和用单一颜色（红色）从暗到亮水平渐变填充的圆的图像，如下图所示：



因为该示例中仅使用一个映射图像和滤镜，所以在 `imageLoadComplete()` 方法中（换句话说，外部图像完成加载时）只创建一次映射图像。通过调用 `MoonSphere` 类的 `createFisheyeMap()` 方法来创建名为 `fisheyeLens` 的置换图图像：

```
var fisheyeLens:BitmapData = createFisheyeMap(radius);
```

在 `createFisheyeMap()` 方法中，在使用 `BitmapData` 类的 `setPixel()` 方法绘制置换图图像时，实际上每次只绘制一个像素。下面列出了 `createFisheyeMap()` 方法的完整代码，并跟有操作方式的逐步说明：

```
private function createFisheyeMap(radius:int):BitmapData
{
    var diameter:int = 2 * radius;

    var result:BitmapData = new BitmapData(diameter,
                                            diameter,
                                            false,
                                            0x808080);

    // Loop through the pixels in the image one by one
    for (var i:int = 0; i < diameter; i++)
    {
        for (var j:int = 0; j < diameter; j++)
        {
            // Calculate the x and y distances of this pixel from
            // the center of the circle (as a percentage of the radius).
            var pctX:Number = (i - radius) / radius;
            var pctY:Number = (j - radius) / radius;

            // Calculate the linear distance of this pixel from
            // the center of the circle (as a percentage of the radius).
            var pctDistance:Number = Math.sqrt(pctX * pctX + pctY * pctY);

            // If the current pixel is inside the circle,
            if (pctDistance <= 1)
                result.setPixel(i, j, 0);
        }
    }
}
```

```

// set its color.
if (pctDistance < 1)
{
    // Calculate the appropriate color depending on the
    // distance of this pixel from the center of the circle.
    var red:int;
    var green:int;
    var blue:int;
    var rgb:uint;
    red = 128 * (1 + 0.75 * pctX * pctX * pctX / (1 - pctY * pctY));
    green = 0;
    blue = 0;
    rgb = (red << 16 | green << 8 | blue);
    // Set the pixel to the calculated color.
    result.setPixel(i, j, rgb);
}
}
return result;
}

```

首先，调用该方法时，将接收参数 `radius`，指示要创建的圆形图像的半径。接下来，该代码将创建 `BitmapData` 对象（将在该对象上绘制圆）。该对象名为 `result`，最终将作为该方法的返回值传递回来。如以下代码片断所示，`result BitmapData` 实例将使用与圆直径相同的宽度和高度创建，且不透明（第三个参数为 `false`），用颜色 `0x808080`（中灰色）预先填充：

```

var result:BitmapData = new BitmapData(diameter,
                                         diameter,
                                         false,
                                         0x808080);

```

接下来，该代码使用两个循环遍历图像的每个像素。外部循环从左到右遍历图像的每一列（使用变量 `i` 表示当前所操作的像素的水平位置），而内部循环从上到下遍历当前列的每个像素（使用变量 `j` 表示当前像素的垂直位置）。下面显示了循环的代码（省略了内部循环的内容）：

```

for (var i:int = 0; i < diameter; i++)
{
    for (var j:int = 0; j < diameter; j++)
    {
        ...
    }
}

```

由于循环逐个遍历像素，因此会为每个像素计算一个值（映射图像中该像素的颜色值）。此过程包含四个步骤：

- 1** 代码计算当前像素沿 x 轴距离圆心的距离 (`i - radius`)。该值除以半径得到半径的百分比而不是绝对距离 (`(i - radius) / radius`)。该百分比值存储在名为 `pctX` 的变量中，而且会计算沿 y 轴的等效值并存储在变量 `pctY`，如以下代码所示：

```

var pctX:Number = (i - radius) / radius;
var pctY:Number = (j - radius) / radius;

```

- 2** 使用标准三角公式（勾股定理），通过 `pctX` 和 `pctY` 计算圆心和当前点之间的直线距离。该值存储在名为 `pctDistance` 的变量中，如下所示：

```

var pctDistance:Number = Math.sqrt(pctX * pctX + pctY * pctY);

```

- 3** 接下来，代码将检查距离百分比是否小于 1（也就是半径的 100%，或者换句话说，所考虑的像素是否处于圆的半径范围之内）。如果该像素落在圆内，将被指定一个计算得到的颜色值（此处省略，在步骤 4 中介绍）；如果不落在圆内，该像素将不发生任何变化，其颜色保留为默认的中灰色：

```

if (pctDistance < 1)
{
    ...
}

```

- 4 对于落在圆内的那些像素，将为其计算一个颜色值。最终颜色将呈现红色色调，范围从圆左边缘的黑(0%)红到圆右边缘的亮(100%)红。颜色值最初按三部分(红、绿和蓝)计算，如下所示：

```
red = 128 * (1 + 0.75 * pctX * pctX * pctX / (1 - pctY * pctY));  
green = 0;  
blue = 0;
```

请注意，实际上只有颜色的红色部分(变量 red)有一个值。为清楚起见，此处显示了绿色和蓝色值(变量 green 和 blue)，不过可以省略。因为该方法的目的是创建一个包含红色渐变的圆，所以不需要绿色和蓝色。

三个颜色值都确定之后，将使用标准移位算法组合成一个整数颜色值，如以下代码所示：

```
rgb = (red << 16 | green << 8 | blue);
```

最后，计算出颜色值后，使用 result BitmapData 对象的 setPixel() 方法将该颜色值实际赋给当前像素，如下所示：

```
result.setPixel(i, j, rgb);
```

第 23 章：在三维 (3D) 环境中工作

3D 基础知识

ActionScript 中的 3D 简介

二维 (2D) 对象和投影在二维屏幕上的三维 (3D) 对象之间的区别在于，三维对象增加了第三维。第三维使对象能够靠近或远离用户的视点。

如果将某个显示对象的 `z` 属性显式设置为数值，则该对象会自动创建一个 3D 转换矩阵。您可以通过更改此矩阵来修改该对象的 3D 转换设置。

此外，3D 旋转与 2D 旋转也有所不同。在 2D 中，旋转轴始终垂直于 `x/y` 平面，即位于 `z` 轴上。在 3D 中，旋转轴可以位于 `x`、`y` 或 `z` 轴中的任一轴上。通过设置显示对象的旋转属性和缩放属性，可以让该对象在 3D 空间中移动。

常见 3D 任务

本章讨论以下与 3D 相关的常见任务：

- 创建 3D 对象
- 在 3D 空间中移动对象
- 在 3D 空间中旋转对象
- 使用透视投影表示深度
- 重新排序显示列表使之与相对 `z` 轴对应，从而让对象以正确的层叠顺序出现在用户面前
- 使用 3D 矩阵转换 3D 对象
- 在 3D 空间中用矢量操作对象
- 使用 `Graphics.drawTriangles()` 方法创建透视
- 使用 UV 映射向 3D 对象添加位图纹理
- 通过设置 `Graphics.drawTriangles()` 方法的剔除参数可加快呈现速度，还可隐藏 3D 对象中背离当前视点的部分。

重要术语和概念

以下参考列表包含将会在本章中遇到的重要术语：

- 透视：在 2D 平面上将平行线表示成聚合于一个消失点，从而获得深度和距离的视觉效果
- 投影：为多维对象生成 2D 图像；3D 投影将 3D 点映射到 2D 平面
- 旋转：通过按圆周运动的方向移动对象内的每个点来更改对象的方向（通常也会更改其位置）
- 转换：通过平移、旋转、缩放、倾斜或这些操作的组合来更改 3D 点或点集
- 平移：通过将对象内的每个点往同一方向移动相同距离来更改对象的位置
- 消失点：在用线性透视法表示时，逐渐远离的平行线看似相交的点
- 矢量：3D 矢量使用笛卡尔坐标 `x`、`y` 和 `z` 表示三维空间中的点或位置
- 顶点：转角点
- 纹理式网格：在 3D 空间中用于定义对象的任意点


- UV 映射: 将纹理或位图应用于 3D 表面的一种方式。UV 映射将值分配给图像上的坐标, 以水平 (U) 轴和垂直 (V) 轴的百分比值形式表示。
- T 值: 当对象靠近或远离当前视点时用于确定 3D 对象大小的缩放系数
- 剔除: 呈现或不呈现具有特定缠绕方式的表面。通过使用剔除, 您可以隐藏对当前视点不可见的表面。

了解 Flash Player 和 AIR 运行时的 3D 功能

在 Flash Player 10 之前的 Flash Player 版本以及 Adobe AIR 1.5 之前的 Adobe AIR 版本中, 显示对象有 x 和 y 两个属性, 用于在 2D 平面上放置显示对象。从 Flash Player 10 和 Adobe AIR 1.5 开始, 每个 ActionScript 显示对象都有一个 z 属性, 通过该属性可以沿 z 轴放置显示对象, z 轴一般用于表示深度或距离。

Flash Player 10 和 Adobe AIR 1.5 引入了对 3D 效果的支持。但是显示对象本质上还是平面的。每个显示对象 (例如 MovieClip 对象或 Sprite 对象) 最终会呈现于二维的单一平面上。通过 3D 功能, 可在三个维中放置、移动、旋转以及按其它方式转换这些平面对象, 还可管理 3D 点以及将这些点转换为 2D x、y 坐标, 这样您就能将 3D 对象投影到 2D 视图上。通过这些功能, 可以模拟出丰富的 3D 效果。

ActionScript 使用的 3D 坐标系与其它坐标系不同。在 ActionScript 中使用 2D 坐标系时, 沿 x 轴向右移动过程中, x 的值增大, 而沿 y 轴向下移动过程中, y 的值增大。3D 坐标系仍遵从这一惯例, 但另外添加了 z 轴, 该轴的值随着远离视点而增大。



ActionScript 3D 坐标系中 x、y 和 z 三条轴的正向。

A. +Z 轴 B. 原点 C. +X 轴 D. +Y 轴

注: 请注意, Flash Player 和 AIR 始终在图层中表示 3D。也就是说, 如果对象 A 在显示列表中位于对象 B 的前面, 则无论这两个对象的 z 轴值为多少, Flash Player 或 AIR 都始终将 A 呈现在 B 的前面。若要解决显示列表顺序与 z 轴顺序之间的冲突, 请使用 `transform.getRelativeMatrix3D()` 方法进行保存, 然后重新排序 3D 显示对象的图层。有关详细信息, 请参阅第 432 页的“[使用 Matrix3D 对象重新排序显示](#)”。

以下 ActionScript 类支持与 3D 相关的新功能:

- 1 `flash.display.DisplayObject` 类包含 z 属性和新的旋转和缩放属性, 这些属性用于操作 3D 空间中的显示对象。
`DisplayObject.local3DToGlobal()` 方法提供了一种将 3D 几何图形投影到 2D 平面的简单方式。
- 2 `flash.geom.Vector3D` 类可用作管理 3D 点的数据结构。该类还支持矢量数学运算。
- 3 `flash.geom.Matrix3D` 类支持复杂的 3D 几何转换, 例如旋转、缩放和平移。

4 flash.geom.PerspectiveProjection 类控制着将 3D 几何图形映射到 2D 视图的相关参数。

在 ActionScript 中，有两种不同的模拟 3D 图像的方式：

- 1 在 3D 空间中排列平面对象并进行动画处理。这种方式需要使用显示对象的 x、y 和 z 属性来对显示对象进行动画处理，或者使用 DisplayObject 类设置旋转和缩放属性。使用 DisplayObject.transform.matrix3D 对象可以实现更为复杂的运动。DisplayObject.transform.perspectiveProjection 对象可自定义显示对象在 3D 透视中的绘制方式。如果需要对主要包含平面的 3D 对象进行动画处理，可以使用这种方式。这种方式的例子包括 3D 图库或者 3D 空间中排列的 2D 动画对象。
- 2 从 3D 几何图形生成 2D 三角形，然后用纹理呈现这些三角形。要使用这种方式，必须首先定义和管理有关 3D 对象的数据，然后将这些数据转换成要呈现的 2D 三角形。可以将位图纹理映射到这些三角形，然后使用 Graphics.drawTriangles() 方法将三角形绘制为图形对象。这种方式的例子包括从文件中加载 3D 模型数据并将模型呈现到屏幕上，或以三角形网格形式生成和绘制 3D 图形。

创建和移动 3D 对象

若要将 2D 显示对象转换为 3D 显示对象，可以将其 z 属性显式设置为一个数值。如果为 z 属性指定一个值，则会为显示对象创建一个新的 Transform 对象。通过设置 DisplayObject.rotationX 或 DisplayObject.rotationY 属性，也可以创建新的 Transform 对象。Transform 对象包含 Matrix3D 属性，该属性控制显示对象在 3D 空间中的表示方式。

下面的代码设置名为“leaf”的显示对象的坐标：

```
leaf.x = 100; leaf.y = 50; leaf.z = -30;
```

在 leaf 的 Transform 对象的 matrix3D 属性中，可以查看这些值，以及从这些值派生出的属性：

```
var leafMatrix:Matrix3D = leaf.transform.matrix3D;  
  
trace(leafMatrix.position.x);  
trace(leafMatrix.position.y);  
trace(leafMatrix.position.z);  
trace(leafMatrix.position.length);  
trace(leafMatrix.position.lengthSquared);
```

有关 Transform 对象的属性的信息，请参阅 [Transform](#) 类。有关 Matrix3D 对象的属性的信息，请参阅 [Matrix3D](#) 类。

在 3D 空间中移动对象

通过更改对象的 x、y 或 z 属性的值，可以在 3D 空间中移动对象。如果更改 z 属性的值，对象看起来就会靠近或远离观察者。

下面的代码在响应事件时，通过更改两个椭圆的 z 属性的值，沿着各自的 z 轴前后移动这两个椭圆。ellipse2 的移动速度比 ellipse1 快：其 z 属性针对每个 Frame 事件增加 20 倍，而 ellipse1 的 z 属性增加 10 倍：

```

var depth:int = 1000;


function ellipse1FrameHandler(e:Event):void
{
    ellipse1Back = setDepth(e, ellipse1Back);
    e.currentTarget.z += ellipse1Back * 10;
}
function ellipse2FrameHandler(e:Event):void
{
    ellipse2Back = setDepth(e, ellipse1Back);
    e.currentTarget.z += ellipse1Back * 20;
}
function setDepth(e:Event, d:int):int
{
    if(e.currentTarget.z > depth)
    {
        e.currentTarget.z = depth;
        d = -1;
    }
    else if (e.currentTarget.z < 0)
    {
        e.currentTarget.z = 0;
        d = 1;
    }
}

```

在 3D 空间中旋转对象

可以通过三种方式旋转对象，具体取决于如何设置对象的旋转属性：rotationX、rotationY 和 rotationZ。

下图显示两个未旋转的正方形：



下一个图显示两个正方形，它们通过增加正方形容器的 rotationY 属性在 y 轴上旋转。通过旋转这两个正方形的容器或父显示对象，可以旋转这两个正方形：


```
container.rotationY += 10;
```



下一个图显示当设置正方形容器的 rotationX 属性时会发生的变化。此操作会在 x 轴上旋转正方形。



下一个图显示当增大正方形容器的 rotationZ 属性时发生的变化。此操作会在 z 轴上旋转正方形。



显示对象可以在 3D 空间中同时移动和旋转。

将 3D 对象投影到 2D 视图上

flash.geom 包中的 [PerspectiveProjection](#) 类提供了一种在 3D 空间中移动显示对象时应用基本透视的简单方式。

如果未显式创建 3D 空间的透视投影，3D 引擎将使用默认的 PerspectiveProjection 对象，该对象存在于根上并会传播到其所有子项上。

用于定义 PerspectiveProjection 对象如何显示 3D 空间的三个属性是：

- fieldOfView
- projectionCenter
- focalLength

修改 fieldOfView 的值将自动修改 focalLength 的值，反之亦然，因为这两个属性相互依赖。

在给出 fieldOfView 值的情况下，用于计算 focalLength 属性的公式为：

```
focalLength = stageWidth/2 * (cos(fieldOfView/2) / sin(fieldOfView/2))
```

通常，您需要显式修改 fieldOfView 属性。

视野

通过操作 PerspectiveProjection 类的 fieldOfView 属性，可以使逐渐靠近观察者的 3D 显示对象变大，而使逐渐远离观察者的对象变小。

fieldOfView 属性指定一个介于 0 到 180 度的角度，该角度确定透视投影的强度。该值越大，沿 z 轴移动的显示对象的扭曲程度就越强。如果 fieldOfView 值较小，则缩放程度较低，从而使对象在空间中只稍稍后移。如果 fieldOfView 值较大，则会导致较大的扭曲，并出现较大的移动。如果达到最大值 180 度，则会出现极端的鱼眼照相机镜头效果。

投影中心

projectionCenter 属性表示透视投影的消失点。该属性作为相对于舞台左上角默认注册点 (0,0) 的偏移量。

当对象逐渐远离观察者时，该对象将朝消失点倾斜直到最终消失。想象一下无限长的走廊。当朝走廊远处看时，两边的墙壁将聚合到走廊远处的消失点。

如果消失点位于舞台的中心，则走廊将消失于该中心点。projectionCenter 属性的默认值是舞台的中心。例如，如果希望元素出现在舞台的左边，而 3D 区域出现在右边，请将 projectionCenter 设置为舞台右边的点，使之成为 3D 查看区域的消失点。

焦距

focalLength 属性表示视点原点 (0,0,0) 与显示点在 z 轴上的位置之间的距离。

较长的焦距相当于视野较窄、对象间距离经过压缩的摄远镜头。较短的焦距相当于广角镜头，可获得较宽的视野和较大的扭曲。中等的焦距相当于肉眼所见的效果。

通常，当显示对象移动时，focalLength 属性会在透视转换过程中动态地重新进行计算，不过您可以显式设置该属性。

默认透视投影值

在根上创建的默认 PerspectiveProjection 对象具有以下值：

- fieldOfView: 55
- perspectiveCenter: stageWidth/2, stageHeight/2
- focalLength: stageWidth/ 2 * (cos(fieldOfView/2) / sin(fieldOfView/2))

如果您没有创建自己的 PerspectiveProjection 对象，则会默认使用这些值。

如果要自行修改 projectionCenter 和 fieldOfView 属性，则可以实例化您自己的 PerspectiveProjection 对象。在这种情况下，新建对象的默认值如下（假设默认舞台大小为 500 x 500）：

- fieldOfView: 55
- perspectiveCenter: 250,250
- focalLength: 480.24554443359375

示例：透视投影

下面的示例演示如何使用透视投影来创建 3D 空间。该示例演示如何通过 projectionCenter 属性来修改消失点和更改空间的透视投影。进行这种修改后，将强制重新计算 focalLength 和 fieldOfView（及其相关的 3D 空间扭曲）。

此示例：

- 1 创建一个名为 center 的 sprite，作为包含十字准线的圆
- 2 将 center sprite 的坐标分配给根的 transform 属性的 perspectiveProjection 属性的 projectionCenter 属性
- 3 为鼠标事件添加事件侦听器，用于调用修改 projectionCenter 的事件处理函数，以便跟踪 center 对象的位置
- 4 创建四个折叠样式的框，这四个框将形成透视空间的墙壁

在测试此示例时，ProjectionDragger.swf 将圆拖动到不同的位置。消失点将跟随圆移动，直到释放圆时才固定下来。在移动投影中心使其远离舞台中心时，观察包围空间的框所发生的拉伸和出现的扭曲。

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。在 Samples/ProjectionDragger 文件夹中可以找到 ProjectionDragger 应用程序文件。

```
package
{
    import flash.display.Sprite;
    import flash.display.Shape;
    import flash.geom.Point;
    import flash.events.*;
    public class ProjectionDragger extends Sprite
    {
        private var center : Sprite;
        private var boxPanel:Shape;
        private var inDrag:Boolean = false;

        public function ProjectionDragger():void
        {
            createBoxes();
            createCenter();
        }
        public function createCenter():void
        {
            var centerRadius:int = 20;

            center = new Sprite();

            // circle
            center.graphics.lineStyle(1, 0x000099);
            center.graphics.beginFill(0xCCCCCC, 0.5);
            center.graphics.drawCircle(0, 0, centerRadius);
            center.graphics.endFill();
            // cross hairs
            center.graphics.moveTo(0, centerRadius);
            center.graphics.lineTo(0, -centerRadius);
            center.graphics.moveTo(centerRadius, 0);
            center.graphics.lineTo(-centerRadius, 0);
            center.x = 175;
            center.y = 175;
            center.z = 0;
            this.addChild(center);

            center.addEventListener(MouseEvent.MOUSE_DOWN, startDragProjectionCenter);
            center.addEventListener(MouseEvent.MOUSE_UP, stopDragProjectionCenter);
            center.addEventListener( MouseEvent.MOUSE_MOVE, doDragProjectionCenter);
            root.transform.perspectiveProjection.projectionCenter = new Point(center.x, center.y);
        }
        public function createBoxes():void
        {
            // createBoxPanel();
            var boxWidth:int = 50;
            var boxHeight:int = 50;
            var numLayers:int = 12;
            var depthPerLayer:int = 50;

            // var boxVec:Vector.<Shape> = new Vector.<Shape>(numLayers);
            for (var i:int = 0; i < numLayers; i++)
            {
                this.addChild(createBox(150, 50, (numLayers - i) * depthPerLayer, boxWidth, boxHeight,
                0xCCCCFF));
                this.addChild(createBox(50, 150, (numLayers - i) * depthPerLayer, boxWidth, boxHeight,
                0xFFCCCC));
                this.addChild(createBox(250, 150, (numLayers - i) * depthPerLayer, boxWidth, boxHeight,
                0xCCFFCC));
                this.addChild(createBox(150, 250, (numLayers - i) * depthPerLayer, boxWidth, boxHeight,
                0xDDDDDD));
            }
        }
    }
}
```

```

        }
    }

    public function createBox(xPos:int = 0, yPos:int = 0, zPos:int = 100, w:int = 50, h:int = 50,
color:int = 0xDDDDDD):Shape
{
    var box:Shape = new Shape();
    box.graphics.lineStyle(2, 0x666666);
    box.graphics.beginFill(color, 1.0);
    box.graphics.drawRect(0, 0, w, h);
    box.graphics.endFill();
    box.x = xPos;
    box.y = yPos;
    box.z = zPos;
    return box;
}

public function startDragProjectionCenter(e:Event)
{
    center.startDrag();
    inDrag = true;
}

public function doDragProjectionCenter(e:Event)
{
    if (inDrag)
    {
        root.transform.perspectiveProjection.projectionCenter = new Point(center.x, center.y);
    }
}

public function stopDragProjectionCenter(e:Event)
{
    center.stopDrag();
    root.transform.perspectiveProjection.projectionCenter = new Point(center.x, center.y);
    inDrag = false;
}
}
}

```

对于更加复杂的透视投影，请使用 Matrix3D 类。

执行复杂的 3D 转换

使用 Matrix3D 类可以转换坐标空间内的 3D 点，也可以将 3D 点从一个坐标空间映射到另一个坐标空间。

不需要理解矩阵运算就可以使用 Matrix3D 类。大多数常见的转换操作都可以通过该类的方法进行处理。您不必担心如何显式设置或计算矩阵中每个元素的值。

将显示对象的 z 属性设置为数值后，可以使用该显示对象的 Transform 对象的 Matrix3D 属性来检索显示对象的转换矩阵：

```
var leafMatrix:Matrix3D = this.transform.matrix3D;
```

您可以用 Matrix3D 对象的方法对显示对象执行平移、旋转、缩放和透视投影。

使用 Vector3D 类及其 x、y 和 z 属性可管理 3D 点。该类还可以表示具有方向和大小的物理空间矢量。通过 Vector3D 类的方法，可以执行有关空间矢量的常见计算，例如加法、点积和叉积计算。

注：Vector3D 类与 ActionScript Vector 类无关。Vector3D 类包含的属性和方法用于定义和操作 3D 点，而 Vector 类则支持类型对象数组。

创建 Matrix3D 对象

有三种创建或检索 Matrix3D 对象的主要方式：

- 1 使用 Matrix3D() 构造函数实例化新的矩阵。Matrix3D() 构造函数需要使用包含 16 个数值的 Vector 对象，并将每个值放入一个矩阵单元中。例如：

```
var rotateMatrix:Matrix3D = new Matrix3D(1,0,0,1, 0,1,0,1, 0,0,1,1, 0,0,0,1);
```
- 2 设置显示对象的 z 属性的值。然后从该对象的 transform.matrix3D 属性检索转换矩阵。
- 3 通过获取根显示对象的 perspectiveProjection.matrix3D 属性的值，检索用于控制舞台上 3D 对象显示方式的 Matrix3D 对象。

应用多种 3D 转换

通过 Matrix3D 对象，可以一次应用多种 3D 转换。例如，如果要旋转、缩放然后移动立方体，可以对立方体的每个点分别应用这三种转换。但是，还有一种高效得多的方法，即在一个 Matrix3D 对象中预先计算多种转换，然后对每个点执行一次矩阵转换。

注：矩阵转换的应用顺序非常重要。矩阵运算的顺序是不能交换的。例如，先应用旋转再应用平移与先应用平移再应用旋转，二者的结果是不同的。

下面的示例演示执行多种 3D 转换的两种方式。

```
package {
    import flash.display.Sprite;
    import flash.display.Shape;
    import flash.display.Graphics;
    import flash.geom.*;
}

public class Matrix3DTransformsExample extends Sprite
{
    private var rect1:Shape;
    private var rect2:Shape;

    public function Matrix3DTransformsExample():void
    {
        var pp:PerspectiveProjection = this.transform.perspectiveProjection;
        pp.projectionCenter = new Point(275,200);
        this.transform.perspectiveProjection = pp;

        rect1 = new Shape();
        rect1.x = -70;
        rect1.y = -40;
        rect1.z = 0;
        rect1.graphics.beginFill(0xFF8800);
        rect1.graphics.drawRect(0,0,50,80);
        rect1.graphics.endFill();
        addChild(rect1);

        rect2 = new Shape();
        rect2.x = 20;
        rect2.y = -40;
        rect2.z = 0;
        rect2.graphics.beginFill(0xFF0088);
        rect2.graphics.drawRect(0,0,50,80);
        rect2.graphics.endFill();
        addChild(rect2);
    }
}
```

```
        doTransforms();
    }

    private function doTransforms():void
    {
        rect1.rotationX = 15;
        rect1.scaleX = 1.2;
        rect1.x += 100;
        rect1.y += 50;
        rect1.rotationZ = 10;

        var matrix:Matrix3D = rect2.transform.matrix3D;
        matrix.appendRotation(15, Vector3D.X_AXIS);
        matrix.appendScale(1.2, 1, 1);
        matrix.appendTranslation(100, 50, 0);
        matrix.appendRotation(10, Vector3D.Z_AXIS);
        rect2.transform.matrix3D = matrix;
    }
}
```

在 `doTransforms()` 方法中，第一个代码块使用 `DisplayObject` 属性更改矩形形状的旋转、缩放和位置。第二个代码块使用 `Matrix3D` 类的方法执行相同的转换。

使用 `Matrix3D` 方法的主要优势在于，所有的计算都是在矩阵中提前执行的。然后，只要设置了显示对象的 `transform.matrix3D` 属性，就只需对显示对象应用一次矩阵转换。通过设置 `DisplayObject` 属性可使源代码更易于阅读。但是，每次设置旋转和缩放属性后，会导致进行大量计算并更改显示对象的多个属性。

如果您的代码要多次对显示对象应用相同的复杂转换，应将 `Matrix3D` 对象保存为变量，然后反复应用该变量。

使用 `Matrix3D` 对象重新排序显示

如前所述，显示列表中显示对象的层叠顺序确定了对象的显示顺序，这些对象的相对 z 轴值对显示顺序没有影响。如果您的动画将显示对象的属性转换为一种与显示列表中的顺序不同的顺序，则观察者看到的显示对象层叠顺序与 z 轴层叠顺序不一致。因此，在视觉上应当远离观察者的对象可能反而会靠近观察者。

为确保 3D 显示对象的层叠顺序对应于对象的相对深度，请使用如下方法：

- 1 使用 `Transform` 对象的 `getRelativeMatrix3D()` 方法获取 3D 子显示对象的相对 z 轴值。
- 2 使用 `removeChild()` 方法从显示列表中删除对象。
- 3 根据显示对象的相对 z 轴值对显示对象进行排序。
- 4 使用 `addChild()` 方法以相反顺序将子对象添加到显示列表中。

这种重新排序方法可确保您的对象按照各自的相对 z 轴值进行显示。

下面的代码将 3D 框的六面设置为正确的显示顺序。这段代码对经过旋转的框的各面进行重新排序：

```

public var faces:Array; . . .

public function ReorderChildren()
{
    for(var ind:uint = 0; ind < 6; ind++)
    {
        faces[ind].z = faces[ind].child.transform.getRelativeMatrix3D(root).position.z;
        this.removeChild(faces[ind].child);
    }
    faces.sortOn("z", Array.NUMERIC | Array.DESCENDING);
    for (ind = 0; ind < 6; ind++)
    {
        this.addChild(faces[ind].child);
    }
}

```

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。在 Samples/ReorderByZ 文件夹中可以找到这些应用程序文件。

通过三角形获得 3D 效果

在 ActionScript 中，可以使用 `Graphics.drawTriangles()` 方法执行位图转换，因为 3D 模型是通过空间中的一组三角形来表示的。（但是，Flash Player 和 AIR 不支持深度缓冲，因此显示对象在本质上仍然是平面的，即 2D。这一点在第 424 页的“[了解 Flash Player 和 AIR 运行时的 3D 功能](#)”中进行了说明。）`Graphics.drawTriangles()` 方法通过一组坐标来绘制三角形路径，类似于 `Graphics.drawPath()` 方法。

若要熟悉 `Graphics.drawPath()` 方法，请参阅第 277 页的“[绘制路径](#)”。

`Graphics.drawTriangles()` 方法使用 `Vector` 对象。`<Number>`，用于指定三角形路径的点位置：

```
drawTriangles(vertices:Vector.<Number>, indices:Vector.<int> = null, uvtData:Vector.<Number> = null,
culling:String = "none") :void
```

`drawTriangles()` 的第一个参数是唯一的必需参数：`vertices` 参数。该参数是由定义坐标的数字组成的矢量，通过该矢量即可绘制三角形。每三组坐标（六个数字）表示一个三角形路径。如果没有 `indices` 参数，矢量的长度应始终为六的倍数，因为每个三角形都需要三对坐标（三组 `x/y` 值对）。例如：

```
graphics.beginFill(0xFF8000);
graphics.drawTriangles(
    Vector.<Number>([
        10,10, 100,10, 10,100,
        110,10, 110,100, 20,100]));

```

这些三角形不共享任何点，但是如果有关共享点，就可以使用第二个 `drawTriangles()` 参数 `indices` 来对多个三角形重复使用 `vertices` 矢量中的值。


使用 `indices` 参数时，请注意，`indices` 值是点索引，而不是直接与 `vertices` 数组元素相关的索引。也就是说，`vertices` 矢量中由 `indices` 定义的索引实际上是除以 2 后的实际索引。例如，对于 `vertices` 矢量的第三个点，即使该点的第一个数值开始于矢量索引 4，也使用 `indices` 值 2。

例如，使用 `indices` 参数合并两个三角形使二者共享对边。

```
graphics.beginFill(0xFF8000);
graphics.drawTriangles(
    Vector.<Number>([10,10, 100,10, 10,100, 100,100]),
    Vector.<int>([0,1,2, 1,3,2]));

```

请注意，尽管正方形现在是通过两个三角形绘制的，但只在 `vertices` 矢量中指定四个点。通过使用 `indices`，两个三角形共享的两点将由每个三角形重复使用。这样可以将顶点总数从 6（12 个数字）减少为 4（8 个数字）：



使用 `vertices` 参数通过两个三角形绘制的正方形

对于较大的三角形网格，这种方法非常有用，因为这种情况下大多数点由多个三角形共享。

所有填充方式都可应用于三角形。填充应用于三角形网格的方式与应用于其它形状的方式是相同的。

转换位图

位图转换可在三维对象上提供透视视觉效果或“纹理”。具体而言，您可以朝消失点的方向扭曲位图，这样图像在朝消失点方向移动时会出现收缩效果。或者，您可以使用二维位图为三维对象创建表面，从而提供纹理视觉效果或将三维对象“包裹”起来。




使用消失点的二维表面和用位图包裹的三维对象。

UV 映射


一旦开始处理纹理，就需要使用 `drawTriangles()` 的 `uvtData` 参数。此参数用于为位图填充设置 UV 映射。

UV 映射是一种纹理化对象的方法。它依赖于两个值：U 水平 (x) 值和 V 垂直 (y) 值。这两个值不是基于像素值，而是基于百分比。0 U 和 0 V 表示图像的坐上角，1 U 和 1 V 表示右下角：




位图图像上的 UV 0 和 1 位置

可以为三角形的矢量指定 UV 坐标，从而将矢量自身关联到图像上的相应位置：



位图图像的三角形区域的 UV 坐标

UV 值与三角形的点保持一致：




三角形的顶点移动且位图发生扭曲，从而使单个点的 UV 值保持不变

由于 ActionScript 3D 转换应用于与位图关联的三角形，位图图像将根据 UV 值应用于三角形。因此，不要使用矩阵运算，而应通过设置或调整 UV 值来实现三维效果。

Graphics.drawTriangles() 方法也接受关于三维转换的一条可选信息：T 值。uvtData 中的 T 值表示 3D 透视，更具体地说，表示相关顶点的缩放系数。UVT 映射向 UV 映射加入了透视修正处理。例如，如果将对象放置在 3D 空间中远离视点的位置，使之显示为原始大小的 50%，则该对象的 T 值为 0.5。因为在 3D 空间中是通过绘制三角形来表示对象的，所以对象在 z 轴上的位置确定对象的 T 值。用于确定 T 值的等式为：

```
T = focalLength / (focalLength + z);
```

在该等式中，`focalLength` 表示焦距或计算得出的“屏幕”位置，它决定了视图中提供的透视大小。



焦距与 z 值

A. 视点 B. 屏幕 C. 3D 对象 D. `focalLength` 值 E. z 值

T 值用于缩放基本形状，使这些形状看起来更远。该值通常用于将 3D 点转换为 2D 点。而对于 UVT 数据，它也用于在透视三角形内的点之间缩放位图。

在定义 UVT 值时，T 值紧跟在为顶点定义的 UV 值后面。加入 T 之后，uvtData 参数 (U、V 和 T) 中的每三个值与 vertices 参数 (x 和 y) 中的每两个值相匹配。在只有 UV 值的情况下，uvtData.length == vertices.length。在加入 T 值的情况下，uvtData.length = 1.5*vertices.length。

下面的示例演示使用 UVT 数据在 3D 空间中旋转平面。本例使用了一幅名为 ocean.jpg 的图像和一个“帮助器”类 ImageLoader，该类用于加载 ocean.jpg 图像以便将其分配给 BitmapData 对象。

下面是 ImageLoader 类的源代码（请将此代码保存到命名为 ImageLoader.as 的文件中）：

```
package {
    import flash.display.*;
    import flash.events.*;
    import flash.net.URLRequest;
    public class ImageLoader extends Sprite {
        public var url:String;
        public var bitmap:Bitmap;
        public function ImageLoader(loc:String = null) {
            if (loc != null){
                url = loc;
                loadImage();
            }
        }
        public function loadImage():void{
            if (url != null){
                var loader:Loader = new Loader();
                loader.contentLoaderInfo.addEventListener(Event.COMPLETE, onComplete);
                loader.contentLoaderInfo.addEventListener(IOErrorEvent.IO_ERROR, onIoError);

                var req:URLRequest = new URLRequest(url);
                loader.load(req);
            }
        }
        private function onComplete(event:Event):void {
            var loader:Loader = Loader(event.target.loader);
            var info:LoaderInfo = LoaderInfo(loader.contentLoaderInfo);
            this.bitmap = info.content as Bitmap;
            this.dispatchEvent(new Event(Event.COMPLETE));
        }
        private function onIoError(event:IOErrorEvent):void {
            trace("onIoError: " + event);
        }
    }
}
```

下面的 ActionScript 用三角形、UV 映射和 T 值使图像获得这样的显示效果：图像朝着消失点逐渐收缩并不断旋转。将此代码保存到命名为 Spinning3dOcean.as 的文件中：

```
package {
    import flash.display.*;
    import flash.events.*;
    import flash.utils.getTimer;

    public class Spinning3dOcean extends Sprite {
        // plane vertex coordinates (and t values)
        var x1:Number = -100,y1:Number = -100,z1:Number = 0,t1:Number = 0;
        var x2:Number = 100,y2:Number = -100,z2:Number = 0,t2:Number = 0;
        var x3:Number = 100,y3:Number = 100,z3:Number = 0,t3:Number = 0;
        var x4:Number = -100,y4:Number = 100,z4:Number = 0,t4:Number = 0;
        var focalLength:Number = 200;
        // 2 triangles for 1 plane, indices will always be the same
        var indices:Vector.<int>;

        var container:Sprite;

        var bitmapData:BitmapData; // texture
        var imageLoader:ImageLoader;
        public function Spinning3dOcean():void {
            indices = new Vector.<int>();
            indices.push(0,1,3, 1,2,3);

            container = new Sprite(); // container to draw triangles in
            container.x = 200;
            container.y = 200;
            addChild(container);

            imageLoader = new ImageLoader("ocean.jpg");
            imageLoader.addEventListener(Event.COMPLETE, onImageLoaded);
        }
        function onImageLoaded(event:Event):void {
            bitmapData = imageLoader.bitmap.bitmapData;
            // animate every frame
            addEventListener(Event.ENTER_FRAME, rotatePlane);
        }
        function rotatePlane(event:Event):void {
            // rotate vertices over time
            var ticker = getTimer()/400;
            z2 = z3 = -(z1 = z4 = 100*Math.sin(ticker));
            x2 = x3 = -(x1 = x4 = 100*Math.cos(ticker));

            // calculate t values
        }
    }
}
```

```
t1 = focalLength/(focalLength + z1);
t2 = focalLength/(focalLength + z2);
t3 = focalLength/(focalLength + z3);
t4 = focalLength/(focalLength + z4);

// determine triangle vertices based on t values
var vertices:Vector.<Number> = new Vector.<Number>();
vertices.push(x1*t1,y1*t1, x2*t2,y2*t2, x3*t3,y3*t3, x4*t4,y4*t4);
// set T values allowing perspective to change
// as each vertex moves around in z space
var uvtData:Vector.<Number> = new Vector.<Number>();
uvtData.push(0,0,t1, 1,0,t2, 1,1,t3, 0,1,t4);

// draw
container.graphics.clear();
container.graphics.beginBitmapFill(bitmapData);
container.graphics.drawTriangles(vertices, indices, uvtData);
}

}


}

}
```

若要测试此示例，请在名为“ocean.jpg”的图像所在的目录中保存这两个类文件。您可以看到原始位图如何转换为在 3D 空间中消失于远处并不断旋转的效果。


剔除

剔除是一个过程，用于确定三维对象的哪些表面因对当前视点不可见而不应被渲染器所呈现。在 3D 空间中，三维对象“背面”的表面对视点不可见：



3D 对象的背面对视点不可见。
A. 视点 **B.** 3D 对象 **C.** 三维对象的背面

从本质上讲，不管三角形的大小、形状或位置如何，所有三角形都会始终呈现。剔除能确保 Flash Player 或 AIR 正确呈现 3D 对象。此外，为了减少呈现周期，有时您会希望渲染器跳过某些三角形。考虑在空间中旋转的立方体。在任何给定时间，您看到的立方体的面数不会超过三个面，这是因为不可见的面朝向立方体背面的其它方向。因为这些面不可见，所以渲染器不应绘制它们。如果不用剔除，Flash Player 或 AIR 就既要绘制正面也要绘制背面。




立方体的某些面对当前视点不可见

因此，Graphics.drawTriangles() 方法采用第四个参数来建立剔除值：

```
public function drawTriangles(vertices:Vector.<Number>, indices:Vector.<int> = null,
uvtData:Vector.<Number> = null, culling:String = "none") : void
```

该剔除参数是来自 `TriangleCulling` 枚举类的值：`TriangleCulling.NONE`、`TriangleCulling.POSITIVE` 和 `TriangleCulling.NEGATIVE`。这些值与定义对象表面的三角形路径的方向有关。用于确定剔除的 ActionScript API 假设 3D 形状的所有外向三角形都是以同一路径方向绘制的。一旦三角形面经过旋转后，其路径方向也会改变。此时可以剔除（不呈现）该三角形。

因此，如果 `TriangleCulling` 值为 `POSITIVE`，则会移除正向路径方向（顺时针）的三角形。如果 `TriangleCulling` 值为 `NEGATIVE`，则会移除负向路径方向（逆时针）的三角形。对于立方体，朝前的表面具有正向路径方向，而朝后的表面具有负向路径方向：



“展开”的立方体，用以显示路径方向。当立方体“恢复原状”后，背面路径方向变为相反方向。

若要了解剔除的工作原理，请从前面第 434 页的“UV 映射”中的示例开始，将 `drawTriangles()` 方法的剔除参数设置为 `TriangleCulling.NEGATIVE`：

```
container.graphics.drawTriangles(vertices, indices, uvtData, TriangleCulling.NEGATIVE);
```

请注意，在对象旋转时，未呈现图像的“背面”。

第 24 章：处理视频

Flash 视频是 Internet 上的一项优秀技术。然而，视频的传统演示形式（在一个矩形屏幕中演示，下方有一个进度栏和一些控制按钮）只是视频的可能使用方式之一。通过 ActionScript，您可以微调和控制视频的加载、演示和播放。

视频基础知识

视频处理简介

Adobe® Flash® Player 和 Adobe® AIR™ 的一个重要功能是可以使用 ActionScript，以操作其它可视内容（如图像、动画、文本等）的方式显示和操作视频信息。

在 Adobe Flash CS4 Professional 中创建 Flash 视频 (FLV) 文件时，您可以选择包含常用播放控件的外观。不过，您不一定要局限于可用的选项。使用 ActionScript 可以精确控制视频的加载、显示和播放，这意味着您可以创建自己的视频播放器外观，也可以按照所需的任何非传统方式使用视频。

在 ActionScript 中使用视频涉及多个类的联合使用：

- **Video 类**：舞台上的实际视频内容框是 Video 类的一个实例。Video 类是一种显示对象，因此可以使用适用于其它显示对象的同样的技术（比如定位、应用变形、应用滤镜和混合模式等）进行操作。
- **NetStream 类**：在加载将由 ActionScript 控制的视频文件时，NetStream 实例表示该视频内容的源，在本例中为视频数据流。使用 NetStream 实例也涉及 NetConnection 对象的使用，该对象是到视频文件的连接，它好比是视频数据馈送的通道。
- **Camera 类**：在通过连接到用户计算机的摄像头处理视频数据时，会使用一个 Camera 实例来表示视频内容的源，即用户的摄像头和它所提供的视频数据。

在加载外部视频时，您可以从标准 Web 服务器加载文件以便进行渐进式下载，也可以使用由专门的服务器（如 Adobe 的 Flash® Media Server）传送的视频流。

常见视频任务

本章介绍了以下您将希望执行的与视频相关的任务：

- 显示和控制屏幕上的视频
- 加载外部视频文件
- 控制视频播放
- 使用全屏
- 处理视频文件中的元数据和提示点信息
- 捕获和显示从用户摄像头输入的视频

重要概念和术语

- 提示点：一个可以放在视频文件内特定时刻的标记，例如，可用作书签以便定位到该时刻或提供与该时刻相关联的其它数据。
- 编码：以一种格式接收视频数据并将其转换为另一种视频数据格式的过程；例如，接收高分辨率的源视频并将其转换为适合于 Internet 传送的格式。

- 帧：单一的一段视频信息；每一帧都类似于一个代表某一时刻的快照的静止图像。通过按顺序高速播放各个帧，可产生动画视觉效果。
- 关键帧：包含帧的完整信息的视频帧。关键帧后面的其它帧仅包含有关它们与关键帧之间的差异的信息，而不包含完整的帧信息。
- 元数据：有关视频文件的信息，嵌入在视频文件中并可在加载视频时检索。
- 渐进式下载：当视频文件从标准 Web 服务器传送时，会使用渐进式下载来加载视频数据，这意味着会按顺序加载视频信息。其好处是不必等待整个文件下载完毕即可开始播放视频；不过，它会阻止您向前跳到视频中尚未加载的部分。
- 流式传输：渐进式下载的一种替代方法，使用流式传输（有时称为“实流”）技术和一台专用视频服务器通过 Internet 传送视频。使用流式传输，用于查看视频的计算机不必一次下载整个视频。为了加快下载速度，在任何时刻，计算机均只需要整个视频信息的一部分。由于使用一台专用服务器来控制视频内容的传送，因此可以在任何时刻访问视频的任何部分，而无需等待其下载完毕后才能进行访问。

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。由于本章介绍的是在 ActionScript 中使用视频，因此，本章中的许多代码清单都涉及处理视频对象（可能是在 Flash 中创建并放置在舞台上的对象，也可能是使用 ActionScript 创建的对象）。测试范例将涉及在 Flash Player 或 AIR 中查看结果，以了解代码对视频的影响。

多数示例代码清单都操作 Video 对象而不显式创建该对象。要测试本章中的这些代码清单，请执行以下操作：

- 1 创建一个空的 Flash 文档。
- 2 在时间轴上选择一关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 如有必要，请打开“库”面板。
- 5 从“库”面板菜单中，选择“新建视频”。
- 6 在“视频属性”对话框中，输入新视频元件的名称，然后在“类型”字段中选择“视频(受 ActionScript 控制)”。单击“确定”创建一个视频元件。
- 7 将视频元件的一个实例从“库”面板拖动到舞台上。
- 8 使视频实例保持选中状态，在“属性”检查器中，为其指定实例名称。该名称应与在示例代码清单中为视频实例使用的名称相匹配，例如，如果代码清单操作名为 vid 的 Video 对象，则应将您的舞台实例也命名为 vid。
- 9 使用“控制”>“测试影片”运行程序。

在屏幕上，您将看到按照代码清单所指定的要求操作视频的结果。

本章中的一些示例代码清单除了包括示例代码以外，还包括类定义。在这些清单中，除了前几步之外，在测试 SWF 之前，还需要创建用在示例中的类。要创建在示例代码清单中定义的类，请执行以下操作：

- 1 请确保已经保存了将用于测试的 FLA 文件。
- 2 从主菜单中选择“文件”>“新建”。
- 3 在“新建文档”对话框的“类型”部分，选择“ActionScript 文件”。单击“确定”创建新的 ActionScript 文件。
- 4 将类定义代码从示例复制到 ActionScript 文档中。
- 5 从主菜单中选择“文件”>“保存”。将该文件保存在 Flash 文档所在的目录中。文件名应与代码清单中的类的名称一致。例如，如果代码清单定义一个名为“VideoTest”的类，则应将 ActionScript 文件保存为“VideoTest.as”。
- 6 返回到 Flash 文档。
- 7 使用“控制”>“测试影片”运行程序。

您将在屏幕上看到示例的结果。

用于测试示例代码清单的其它方法在第 29 页的“[测试本章内的示例代码清单](#)”中有更详细的介绍。

了解视频格式

除了 Adobe FLV 视频格式之外，Flash Player 和 Adobe AIR 还支持从 MPEG-4 标准文件格式中以 H.264 和 HE-AAC 编码的视频和音频。这些格式以更低的比特率提供高质量的视频流。开发人员可利用业界标准工具（包括 Adobe Premiere Pro 和 Adobe After Effects）来创建并提供引入注目的视频内容。

类型	格式	容器
视频	H.264	MPEG-4: MP4、M4V、F4V、3GPP
视频	FLV 文件	Sorenson Spark
视频	FLV 文件	ON2 VP6
音频	AAC+ / HE-AAC / AAC v1 / AAC v2	MPEG-4: MP4、M4V、F4V、3GPP
音频	Mp3	Mp3
音频	Nellymoser	FLV 文件
音频	Speex	FLV 文件

Flash Player 和 AIR 与编码的视频文件的兼容性

Flash Player 7 支持用 Sorenson™ Spark™ 视频编解码器编码的 FLV 文件。Flash Player 8 支持用 Flash Professional 8 中的 Sorenson Spark 或 On2 VP6 编码器编码的 FLV 文件。On2 VP6 视频编解码器支持 Alpha 通道。

Flash Player 9.0.115.0 及更高版本支持从标准 MPEG-4 容器格式派生的文件。这些文件包括 F4V、MP4、M4A、MOV、MP4V、3GP 和 3G2（如果这些文件包含 H.264 视频和 / 或 HEAAC v2 编码音频）。与 Sorenson 或 On2 中相同的编码配置文件相比，H.264 可以在更低的比特率下传送更高品质的视频。HE-AAC v2 是 AAC 的扩展，AAC 是在 MPEG-4 视频标准中定义的一种标准音频格式。HE-AAC v2 使用频带复制 (SBR) 和参量立体声 (PS) 技术来提高低比特率下的编码效率。

下表列出了支持的编解码器。表中还显示了相应的 SWF 文件格式以及播放这些文件所需的 Flash Player 和 AIR 版本：

编解码器	SWF 文件格式版本（支持的最早发布版本）	Flash Player 和 AIR（播放所需要的最早的版本）
Sorenson Spark	6	Flash Player 6、Flash Lite 3
On2 VP6	6	Flash Player 8、Flash Lite 3。 只有 Flash Player 8 及更高版本才支持 On2 VP6 视频的发布和播放。
H.264 (MPEG-4 Part 10)	9	Flash Player 9 Update 3、AIR 1.0
ADPCM	6	Flash Player 6、Flash Lite 3
Mp3	6	Flash Player 6、Flash Lite 3
AAC (MPEG-4 Part 3)	9	Flash Player 9 Update 3、AIR 1.0
Speex (音频)	10	Flash Player 10、AIR 1.5
Nellymoser	6	Flash Player 6

了解 Adobe F4V 和 FLV 视频文件格式

Adobe 提供了 F4V 和 FLV 视频文件格式，以用于向 Flash Player 和 AIR 流式传输内容。有关这些视频文件格式的完整描述，请参阅 www.adobe.com/go/video_file_format_cn。

F4V 视频文件格式

从 Flash Player Update 3 (9.0.115.0) 和 AIR 1.0 开始，Flash Player 和 AIR 支持 Adobe F4V 视频格式，该格式基于 ISO MP4 格式，MP4 格式的不同子集支持不同的功能。Flash Player 需要有效的 F4V 文件来从以下某个顶级框开始：

- ftyp

ftyp 框标识程序为播放特定的文件格式而必须支持的功能。

- moov

moov 框实际是 F4V 文件的标头。该框包含一个或多个其它框，这些框又包含定义 F4V 数据结构的其它框。F4V 文件必须包含且只能包含一个 moov 框。

- mdat

mdat 框包含 F4V 文件的数据负载。一个 FV 文件只包含一个 mdat 框。moov 框也必须存在于文件中，因为 mdat 框在单独使用时没有意义。

F4V 文件支持 big-endian 字节顺序的多字节整数，按照该顺序，最高有效字节在最低地址中最先出现。

FLV 视频文件格式

Adobe FLV 文件格式包含可供 Flash Player 传送的音频和视频编码数据。可以使用编码器（如 Adobe Media Encoder 或 Sorenson™ Squeeze）将 QuickTime 或 Windows Media 视频文件转换为 FLV 文件。

注：可通过将视频导入 Flash，然后再导出为 FLV 文件来创建 FLV 文件。可以使用“FLV 导出”插件从受支持的视频编辑应用程序中导出 FLV 文件。若要从 Web 服务器加载 FLV 文件，请向您的 Web 服务器注册文件扩展名和 MIME 类型。请查看您的 Web 服务器文档。FLV 文件的 MIME 类型是 video/x-flv。有关详细信息，请参阅第 467 页的“[关于配置 FLV 文件以便在服务器上托管](#)”。

有关 FLV 文件的详细信息，请参阅 第 467 页的“[FLV 文件的高级主题](#)”。

外部视频和嵌入视频

使用外部视频文件可以提供使用导入的视频时不可用的某些功能：

- 可在应用程序中使用较长的视频剪辑，而不会降低播放速度。外部视频文件可使用缓存内存，这意味着大文件将分成小片断存储，并可以动态访问。因此，外部 F4V 和 FLV 文件所需要的内存比嵌入的视频文件要少。
- 外部视频文件的帧速率可以不同于它所播放的 SWF 文件。例如，可以将 SWF 文件帧速率设置为 30 帧 / 秒 (fps)，而将视频帧速率设置为 21 fps。与嵌入的视频相比，此项设置可使您更好地控制视频，确保视频顺畅地播放。此项设置还允许您以不同的帧速率播放视频文件，而无需更改现有 SWF 文件的内容。
- 如果使用外部视频文件，则不会在加载视频文件时中断 SWF 内容的播放。导入的视频文件有时可能需要中断文档播放来执行某些功能，例如，访问 CD-ROM 驱动器。视频文件可独立于 SWF 内容执行功能，而不会中断播放。
- 对于外部 FLV 文件，为视频内容加字幕更加简单，这是因为您可以使用事件处理函数访问视频元数据。

了解 Video 类

使用 **Video** 类可以直接在应用程序中显示实时视频流，而无需将其嵌入 SWF 文件中。可以使用 `Camera.getCamera()` 方法捕获并播放实时视频。还可以使用 **Video** 类通过 HTTP 或在本地文件系统中播放视频文件。在项目中使用 **Video** 有多种不同方法：

- 使用 `NetConnection` 和 `NetStream` 类动态加载视频文件并在 `Video` 对象中显示视频。
- 从用户摄像头捕获输入。有关详细信息，请参阅第 462 页的“[捕获摄像头输入](#)”。
- 使用 `FLVPlayback` 组件。

注：舞台上 `Video` 对象的实例是 `Video` 类的实例。

尽管 `Video` 类位于 `flash.media` 包中，但它继承自 `flash.display.DisplayObject` 类。因此，所有显示对象功能（如矩阵转换和滤镜）也适用于 `Video` 实例。

有关详细信息，请参阅第 241 页的“[处理显示对象](#)”、第 283 页的“[处理几何结构](#)”和第 293 页的“[过滤显示对象](#)”。

加载视频文件

使用 `NetStream` 和 `NetConnection` 类加载视频是一个多步骤过程：

- 1 创建一个 `NetConnection` 对象。如果要连接到本地视频文件或者未使用 `Adobe Flash Media Server 2` 之类的服务器的视频文件，请将 `null` 传给 `connect()` 方法，以从 HTTP 地址或本地驱动器上播放视频文件。如果要连接到服务器，请将该参数设置为包含服务器上视频文件的应用程序的 URI。

```
var nc:NetConnection = new NetConnection();
nc.connect(null);
```

- 2 创建一个接受 `NetConnection` 对象作为参数的 `NetStream` 对象，并指定要加载的视频文件：以下代码片断将 `NetStream` 对象连接到指定的 `NetConnection` 实例，并加载 SWF 文件所在目录中名为 `video.mp4` 的视频文件：

```
var ns:NetStream = new NetStream(nc);
ns.addEventListener(AsyncErrorEvent.ASYNC_ERROR, asyncErrorHandler);
ns.play("video.mp4");
function asyncErrorHandler(event:AsyncErrorEvent):void
{
    // ignore error
}
```

- 3 创建一个新的 `Video` 对象，并使用 `Video` 类的 `attachNetStream()` 方法连接先前创建的 `NetStream` 对象。然后可以使用 `addChild()` 方法将该视频对象添加到显示列表中，如以下代码片断所示：

```
var vid:Video = new Video();
vid.attachNetStream(ns);
addChild(vid);
```

当 Flash Player 执行此代码时，它将尝试从 SWF 文件所在的目录中加载 `video.mp4` 视频文件。

控制视频播放

`NetStream` 类提供了四个用于控制视频播放的主要方法：

`pause()`：暂停视频流的播放。如果视频已经暂停，则调用此方法将不会执行任何操作。

`resume()`：恢复播放暂停的视频流。如果视频已在播放，则调用此方法将不会执行任何操作。

seek(): 搜寻最接近于指定位置（从流的开始位置算起的偏移量，以秒为单位）的关键帧。

togglePause(): 暂停或恢复播放流。

注: 没有 **stop()** 方法。为了停止视频流，必须暂停播放并找到视频流的开始位置。

注: **play()** 方法不会恢复播放，它用于加载视频文件。

以下示例演示如何使用多个不同的按钮控制视频。若要运行下面的示例，请创建一个新文档，并在工作区中添加 4 个按钮实例（**pauseBtn**、**playBtn**、**stopBtn** 和 **togglePauseBtn**）：

```
var nc:NetConnection = new NetConnection();
nc.connect(null);

var ns:NetStream = new NetStream(nc);
ns.addEventListener(AsyncErrorEvent.ASYNC_ERROR, asyncErrorHandler);
ns.play("video.flv");
function asyncErrorHandler(event:AsyncErrorEvent):void
{
    // ignore error
}

var vid:Video = new Video();
vid.attachNetStream(ns);
addChild(vid);

pauseBtn.addEventListener(MouseEvent.CLICK, pauseHandler);
playBtn.addEventListener(MouseEvent.CLICK, playHandler);
stopBtn.addEventListener(MouseEvent.CLICK, stopHandler);
togglePauseBtn.addEventListener(MouseEvent.CLICK, togglePauseHandler);

function pauseHandler(event:MouseEvent):void
{
    ns.pause();
}
function playHandler(event:MouseEvent):void
{
    ns.resume();
}
function stopHandler(event:MouseEvent):void
{
    // Pause the stream and move the playhead back to
    // the beginning of the stream.
    ns.pause();
    ns.seek(0);
}
function togglePauseHandler(event:MouseEvent):void
{
    ns.togglePause();
}
```

播放视频的同时单击 **pauseBtn** 按钮实例会导致视频文件暂停。如果视频已经暂停，则单击此按钮不会执行任何操作。如果之前暂停了播放，则单击 **playBtn** 按钮实例会恢复视频播放；如果视频已在播放，则单击该按钮不会执行任何操作。

检测视频流的末尾

为了侦听视频流的开始和末尾，需要向 **NetStream** 实例添加一个事件侦听器以侦听 **netStatus** 事件。以下代码演示如何在视频播放过程中侦听不同代码：

```
ns.addEventListener(NetStatusEvent.NET_STATUS, statusHandler);
function statusHandler(event:NetStatusEvent):void
{
    trace(event.info.code)
}
```

上面这段代码的输出如下：

```
NetStream.Play.Start
NetStream.Buffer.Empty
NetStream.Buffer.Full
NetStream.Buffer.Empty
NetStream.Buffer.Full
NetStream.Buffer.Empty
NetStream.Buffer.Full
NetStream.Buffer.Flush
NetStream.Play.Stop
NetStream.Buffer.Empty
NetStream.Buffer.Flush
```

您要专门侦听的两段代码为“NetStream.Play.Start”和“NetStream.Play.Stop”，它们会在视频播放的开始和末尾发出信号。

下面的代码片段使用 switch 语句来过滤这两段代码并输出一条消息：

```
function statusHandler(event:NetStatusEvent):void
{
    switch (event.info.code)
    {
        case "NetStream.Play.Start":
            trace("Start [" + ns.time.toFixed(3) + " seconds]");
            break;
        case "NetStream.Play.Stop":
            trace("Stop [" + ns.time.toFixed(3) + " seconds]");
            break;
    }
}
```

通过侦听 netStatus 事件 (NetStatusEvent.NET_STATUS)，您可以生成一个视频播放器，它在当前视频完成播放后加载播放列表中的下一个视频。

在全屏模式下播放视频

Flash Player 和 AIR 使您可以创建全屏应用程序来播放您的视频，并支持将视频放大至全屏。

对于以全屏模式运行的 AIR 内容，在播放过程中将禁用系统屏幕保护程序和节能选项，直至视频输入停止或用户退出全屏模式。

有关使用全屏模式的完整详细信息，请参阅第 237 页的“[处理全屏模式](#)”。

在浏览器中为 **Flash Player** 启用全屏模式

应先通过应用程序的“发布”模板启用全屏模式，然后才能在浏览器中实现 Flash Player 的全屏模式。允许全屏的模板包含 <object> 和 <embed> 标签，这些标签包含 allowFullScreen 参数。下面的示例显示了 <embed> 标签中的 allowFullScreen 参数。

```

<object classid="clsid:D27CDB6E-AE6D-11cf-96B8-444553540000"
       id="fullScreen" width="100%" height="100%"
       codebase="http://fpdownload.macromedia.com/get/flashplayer/current/swflash.cab">
  ...
  <param name="allowFullScreen" value="true" />
  <embed src="fullScreen.swf" allowFullScreen="true" quality="high" bgcolor="#869ca7"
         width="100%" height="100%" name="fullScreen" align="middle"
         play="true"
         loop="false"
         quality="high"
         allowScriptAccess="sameDomain"
         type="application/x-shockwave-flash"
         pluginspage="http://www.adobe.com/go/getflashplayer">
  </embed>
  ...
</object>
```

在 Flash 中，选择“文件”->“发布设置”，然后在“发布设置”对话框中的“HTML”选项卡上，选择“仅 Flash - 允许全屏”模板。

在 Flex 中，确保 HTML 模板包含支持全屏的 `<object>` 和 `<embed>` 标签。

启动全屏模式

对于运行于浏览器中的 Flash Player 内容，可在响应鼠标单击或按键操作时为视频启动全屏模式。例如，您可以在用户单击标签为“全屏”的按钮或者从上下文菜单中选择“全屏”命令时，启动全屏模式。若要响应用户，请将一个事件侦听器添加到发生操作的对象。下面的代码将一个事件侦听器添加到用户单击后可进入全屏模式的按钮：

```

var fullScreenButton:Button = new Button();
fullScreenButton.label = "Full Screen";
addChild(fullScreenButton);
fullScreenButton.addEventListener(MouseEvent.CLICK, fullScreenButtonHandler);

function fullScreenButtonHandler(event:MouseEvent)
{
    stage.displayState = StageDisplayState.FULL_SCREEN;
}
```

代码通过将 `Stage.displayState` 属性设置为 `StageDisplayState.FULL_SCREEN` 来启动全屏模式。这段代码将整个舞台放大为全屏，同时其中的视频根据它在舞台中所占空间的比例一同放大。

`fullScreenSourceRect` 属性可用于指定舞台上要放大为全屏的某块特定区域。首先，定义要放大为全屏的矩形。然后将其赋给 `Stage.fullScreenSourceRect` 属性。此版本的 `fullScreenButtonHandler()` 函数添加了另外两行代码，这些代码只是将视频放大为全屏。

```

private function fullScreenButtonHandler(event:MouseEvent)
{
    var screenRectangle:Rectangle = new Rectangle(video.x, video.y, video.width, video.height);
    stage.fullScreenSourceRect = screenRectangle;
    stage.displayState = StageDisplayState.FULL_SCREEN;
}
```

虽然此示例中为响应鼠标单击而调用了一个事件处理函数，但进入全屏模式的方法对于 Flash Player 和 AIR 都是相同的。定义要缩放的矩形，然后设置 `Stage.displayState` 属性。有关详细信息，请参阅 [ActionScript 3.0 语言和组件参考](#)。

完整的示例（如下所示）添加了额外的代码，这些代码创建连接以及视频的 `NetStream` 对象，并开始播放该视频。

```
package
{
    import flash.net.NetConnection;
    import flash.net.NetStream;
    import flash.media.Video;
    import flash.display.StageDisplayState;
    import fl.controls.Button;
    import flash.display.Sprite;
    import flash.events.MouseEvent;
    import flash.events.FullScreenEvent;
    import flash.geom.Rectangle;

    public class FullScreenVideoExample extends Sprite
    {
        var fullScreenButton:Button = new Button();
        var video:Video = new Video();

        public function FullScreenVideoExample()
        {
            var videoConnection:NetConnection = new NetConnection();
            videoConnection.connect(null);

            var videoStream:NetStream = new NetStream(videoConnection);
            videoStream.client = this;

            addChild(video);

            video.attachNetStream(videoStream);

            videoStream.play("http://www.helpexamples.com/flash/video/water.flv");

            fullScreenButton.x = 100;
            fullScreenButton.y = 270;
            fullScreenButton.label = "Full Screen";
            addChild(fullScreenButton);
            fullScreenButton.addEventListener(MouseEvent.CLICK, fullScreenButtonHandler);
        }

        private function fullScreenButtonHandler(event:MouseEvent)
        {
            var screenRectangle:Rectangle = new Rectangle(video.x, video.y, video.width, video.height);
            stage.fullScreenSourceRect = screenRectangle;
            stage.displayState = StageDisplayState.FULL_SCREEN;
        }

        public function onMetaData(infoObject:Object):void
        {
            // stub for callback function
        }
    }
}
```

`onMetaData()` 函数是处理视频元数据（如果存在）的回调函数。回调函数是运行时响应某种类型的事件而调用的函数。此示例中，`onMetaData()` 函数是满足要求以提供该函数的存根。有关详细信息，请参阅第 450 页的“[编写元数据和提示点的回调方法](#)”。

退出全屏模式

用户通过输入某个快捷键（如 Esc 键）即可退出全屏模式。可以用 ActionScript 终止全屏模式，方法是将 `Stage.displayState` 属性设置为 `StageDisplayState.NORMAL`。下例中的代码在发生 `NetStream.Play.Stop` `netStatus` 事件时终止全屏模式。

```
videoStream.addEventListener(NetStatusEvent.NET_STATUS, netStatusHandler);

private function netStatusHandler(event:NetStatusEvent)
{
    if(event.info.code == "NetStream.Play.Stop")
        stage.displayState = StageDisplayState.NORMAL;
}
```

全屏硬件加速

在设置 `Stage.fullScreenSourceRect` 属性以将舞台的某个矩形部分放大为全屏模式时，如果硬件加速可用并已启用，则 Flash Player 或 AIR 将使用硬件加速。运行时使用计算机上的视频适配器加速将视频或舞台的某个局部放大为全屏大小。

有关全屏模式下硬件加速的详细信息，请参阅第 237 页的“[处理全屏模式](#)”。

流式传输视频文件

若要流式传输 Flash Media Server 中的文件，可以使用 `NetConnection` 和 `NetStream` 类连接到远程服务器实例并播放指定的流。若要指定实时消息传递协议 (RTMP) 服务器，请向 `NetConnection.connect()` 方法传递所需的 RTMP URL（例如“`rtmp://localhost/appName/appInstance`”），而不是传递 `null`。若要播放指定 Flash Media Server 中的指定实时流或录制流，请为 `NetStream.play()` 方法传递一个由 `NetStream.publish()` 发布的实时数据的标识名称，或一个要播放的录制文件名称。有关详细信息，请参阅 Flash Media Server 文档。

了解提示点

可在编码期间将提示点嵌入 Adobe F4V 或 FLV 视频文件。过去，在影片中嵌入提示点是为了给放映员提供了一个可视信号，以指出胶片盘中的胶片即将放完。在 Adobe F4V 和 FLV 视频格式中，提示点的作用在于：当视频流中出现提示点时，在应用程序中触发一个或多个其它动作。

您可以对 Flash 视频使用几种不同类型的提示点。可以使用 ActionScript 与在创建视频文件时嵌入其中的提示点进行交互。

- **导航提示点：**您可以在编码视频文件时，将导航提示点嵌入到视频流和元数据包中。使用导航提示点可以使用户搜索到文件的指定部分。
- **事件提示点：**您可以在编码视频文件时，将事件提示点嵌入到视频流和元数据包中。还可以编写代码来处理视频播放期间在指定点上触发的事件。
- **ActionScript 提示点：**ActionScript 提示点只对 Flash FLVPlayback 组件有用。ActionScript 提示点是您使用 ActionScript 代码创建和访问的外部提示点。您可以编写代码来触发这些与视频播放有关的提示点。这些提示点的精确度要低于嵌入的提示点（最高时相差 1/10 秒），因为视频播放器单独跟踪这些提示点。如果您计划创建一个应用程序，希望用户能在其中导航至提示点，则应在编码文件时创建并嵌入提示点，而不应使用 ActionScript 提示点。您应将提示点嵌入 FLV 文件中，因为这些提示点更加精确。

由于导航提示点会在指定的提示点位置创建一个关键帧，因此可以使用代码将视频播放器的播放头移动到该位置。您可以在视频文件中设置一些希望用户搜索的特定点。例如，视频可能会具有多个章节或段，在这种情况下您就可以在视频文件中嵌入导航提示点，以此方式来控制视频。

有关使用提示点对 Adobe 视频文件进行编码的详细信息，请参阅《使用 Flash》中的“[嵌入提示点](#)”。

您可以通过编写 ActionScript 来访问提示点参数。提示点参数是由回调处理函数接收的事件对象的一部分。

若要在 FLV 文件到达特定提示点时在代码中触发特定动作，请使用 `NetStream.onCuePoint` 事件处理函数。

若要同步对 F4V 视频文件中某个提示点的操作，必须从 `onMetaData()` 或 `onXMPData()` 回调函数检索提示点数据，并使用 ActionScript 3.0 中的 `Time` 类触发提示点。有关 F4V 提示点的详细信息，请参阅第 459 页的“[使用 onXMPData\(\)](#)”。

有关处理提示点和元数据的详细信息，请参阅第 450 页的“[编写元数据和提示点的回调方法](#)”。

编写元数据和提示点的回调方法

当播放器收到特定元数据或到达特定提示点时，可以在应用程序中触发动作。当这些事件发生时，必须将特定回调方法用作事件处理函数。`NetStream` 类指定了在播放期间可发生的以下元数据事件：`onCuePoint`（仅限 FLV 文件）、`onImageData`、`onMetaData`、`onPlayStatus`、`onTextData` 和 `onXMPData`。

必须为这些处理函数编写回调方法，否则，Flash Player 可能会引发错误。例如，以下代码播放 SWF 文件所在文件夹中名为 `video.flv` 的 FLV 文件：

```
var nc:NetConnection = new NetConnection();
nc.connect(null);

var ns:NetStream = new NetStream(nc);
ns.addEventListener(AsyncErrorEvent.ASYNC_ERROR, asyncErrorHandler);
ns.play("video.flv");
function asyncErrorHandler(event:AsyncErrorEvent):void
{
    trace(event.text);
}

var vid:Video = new Video();
vid.attachNetStream(ns);
addChild(vid);
```

上面的代码加载一个名为 `video.flv` 的本地视频文件并侦听要调度的 `asyncError` (`AsyncErrorEvent.ASYNC_ERROR`)。当本机异步代码中引发异常时调度此事件。在本例中，当视频文件中包含元数据或提示点信息，并且未定义相应的侦听器时，将调度此事件。如果您对视频文件的元数据或提示点信息不感兴趣，则可以使用上面的代码处理 `asyncError` 事件并忽略错误。如果 FLV 中有元数据和多个提示点，则 `trace()` 函数将显示以下错误消息：

```
Error #2095: flash.net.NetStream was unable to invoke callback onMetaData.
Error #2095: flash.net.NetStream was unable to invoke callback onCuePoint.
Error #2095: flash.net.NetStream was unable to invoke callback onCuePoint.
Error #2095: flash.net.NetStream was unable to invoke callback onCuePoint.
```

发生错误的原因是 `NetStream` 对象找不到 `onMetaData` 或 `onCuePoint` 回调方法。在应用程序中定义这些回调方法有多种方式：

将 `NetStream` 对象的 `client` 属性设置为一个 `Object`

通过将 `client` 属性设置为一个 `Object` 或设置为 `NetStream` 的一个子类，可以重新发送 `onMetaData` 和 `onCuePoint` 回调方法或彻底忽略这些方法。以下示例演示如何使用空的 `Object` 忽略这些回调方法而不侦听 `asyncError` 事件：

```
var nc:NetConnection = new NetConnection();
nc.connect(null);

var customClient:Object = new Object();

var ns:NetStream = new NetStream(nc);
ns.client = customClient;
ns.play("video.flv");

var vid:Video = new Video();
vid.attachNetStream(ns);
addChild(vid);
```

如果想要侦听 `onMetaData` 或 `onCuePoint` 回调方法，则需要定义用于处理这些回调方法的方法，如以下代码片断所示：

```

var customClient:Object = new Object();
customClient.onMetaData = metaDataHandler;
function metaDataHandler(infoObject:Object):void
{
    trace("metadata");
}

```

上面的代码侦听 `onMetaData` 回调方法并调用 `metaDataHandler()` 方法，后者会输出一个字符串。如果 Flash Player 遇到一个提示点，那么即使未定义 `onCuePoint` 回调方法，也不会生成错误。

创建自定义类并定义用于处理回调方法的方法

以下代码将 `NetStream` 对象的 `client` 属性设置为一个自定义类 `CustomClient`，该类为回调方法定义处理函数：

```

var nc:NetConnection = new NetConnection();
nc.connect(null);

var ns:NetStream = new NetStream(nc);
ns.client = new CustomClient();
ns.play("video.flv");

var vid:Video = new Video();
vid.attachNetStream(ns);
addChild(vid);

```

`CustomClient` 类如下所示：

```

package
{
    public class CustomClient
    {
        public function onMetaData(infoObject:Object):void
        {
            trace("metadata");
        }
    }
}

```

`CustomClient` 类为 `onMetaData` 回调处理函数定义一个处理函数。如果遇到了提示点，并且调用了 `onCuePoint` 回调处理函数，则会调度一个 `asyncError` 事件 (`AsyncErrorEvent.ASYNC_ERROR`)，显示“`flash.net.NetStream` 无法调用回调 `onCuePoint`”。为了防止发生此错误，需要在 `CustomClient` 类中定义一个 `onCuePoint` 回调方法，或者为 `asyncError` 事件定义一个事件处理函数。

扩展 `NetStream` 类并添加处理回调方法的方法

以下代码创建 `CustomNetStream` 类的一个实例，`CustomNetStream` 类在后面的代码清单中定义：

```

var ns:CustomNetStream = new CustomNetStream();
ns.play("video.flv");

var vid:Video = new Video();
vid.attachNetStream(ns);
addChild(vid);

```

以下代码清单定义 `CustomNetStream` 类，该类扩展 `NetStream` 类、处理必要的 `NetConnection` 对象的创建，并处理 `onMetaData` 和 `onCuePoint` 回调处理函数方法：

```
package
{
    import flash.net.NetConnection;
    import flash.net.NetStream;
    public class CustomNetStream extends NetStream
    {
        private var nc:NetConnection;
        public function CustomNetStream()
        {
            nc = new NetConnection();
            nc.connect(null);
            super(nc);
        }
        public function onMetaData(infoObject:Object):void
        {
            trace("metadata");
        }
        public function onCuePoint(infoObject:Object):void
        {
            trace("cue point");
        }
    }
}
```

如果要重命名 CustomNetStream 类中的 onMetaData() 和 onCuePoint() 方法，可以使用以下代码：

```
package
{
    import flash.net.NetConnection;
    import flash.net.NetStream;
    public class CustomNetStream extends NetStream
    {
        private var nc:NetConnection;
        public var onMetaData:Function;
        public var onCuePoint:Function;
        public function CustomNetStream()
        {
            onMetaData = metaDataHandler;
            onCuePoint = cuePointHandler;
            nc = new NetConnection();
            nc.connect(null);
            super(nc);
        }
        private function metaDataHandler(infoObject:Object):void
        {
            trace("metadata");
        }
        private function cuePointHandler(infoObject:Object):void
        {
            trace("cue point");
        }
    }
}
```

扩展 NetStream 类并使其为动态类

可以扩展 NetStream 类并使其子类为动态类，以便可以动态添加 onCuePoint 和 onMetaData 回调处理函数。以下代码清单演示了这一过程：

```

var ns:DynamicCustomNetStream = new DynamicCustomNetStream();
ns.play("video.flv");

var vid:Video = new Video();
vid.attachNetStream(ns);
addChild(vid);

```

DynamicCustomNetStream 类如下所示：

```

package
{
    import flash.net.NetConnection;
    import flash.net.NetStream;
    public dynamic class DynamicCustomNetStream extends NetStream
    {
        private var nc:NetConnection;
        public function DynamicCustomNetStream()
        {
            nc = new NetConnection();
            nc.connect(null);
            super(nc);
        }
    }
}

```

由于 DynamicCustomNetStream 类为动态类，因此，即使 onMetaData 和 onCuePoint 回调处理函数没有处理函数，也不会引发错误。如果想要为 onMetaData 和 onCuePoint 回调处理函数定义方法，可以使用以下代码：

```

var ns:DynamicCustomNetStream = new DynamicCustomNetStream();
ns.onMetaData = metaDataHandler;
ns.onCuePoint = cuePointHandler;
ns.play("http://www.helpexamples.com/flash/video/cuepoints.flv");

var vid:Video = new Video();
vid.attachNetStream(ns);
addChild(vid);

function metaDataHandler(infoObject:Object):void
{
    trace("metadata");
}
function cuePointHandler(infoObject:Object):void
{
    trace("cue point");
}

```

将 NetStream 对象的 client 属性设置为 this

通过将 client 属性设置为 this，应用程序将在当前作用域内查找 onMetaData() 和 onCuePoint() 方法。以下示例演示了这一效果：

```

var nc:NetConnection = new NetConnection();
nc.connect(null);

var ns:NetStream = new NetStream(nc);
ns.client = this;
ns.play("video.flv");

var vid:Video = new Video();
vid.attachNetStream(ns);
addChild(vid);

```

如果调用 `onMetaData` 或 `onCuePoint` 回调处理函数，在不存在处理该回调的方法时不会生成错误。若要处理这些回调处理函数，请在代码中创建 `onMetaData()` 和 `onCuePoint()` 方法，如以下代码片断所示：

```
function onMetaData(infoObject:Object) :void
{
    trace("metadata");
}
function onCuePoint(infoObject:Object) :void
{
    trace("cue point");
}
```

使用提示点和元数据

使用 `NetStream` 回调方法可在视频播放时捕获并处理提示点和元数据事件。

使用提示点

下表描述了可在 Flash Player 和 AIR 中用来捕获 F4V 和 FLV 提示点的回调方法。

运行时	F4V	FLV
Flash Player 9/ AIR1.0		OnCuePoint
		OnMetaData
Flash Player 10		OnCuePoint
	OnMetaData	OnMetaData
	OnXMPData	OnXMPData

下面的示例使用简单的 `for..in` 循环来遍历 `onCuePoint()` 函数收到的 `infoObject` 参数中的每一个属性。它在收到提示点数据时，调用 `trace()` 函数来显示消息：

```
var nc:NetConnection = new NetConnection();
nc.connect(null);

var ns:NetStream = new NetStream(nc);
ns.client = this;
ns.play("video.flv");

var vid:Video = new Video();
vid.attachNetStream(ns);
addChild(vid);

function onCuePoint(infoObject:Object) :void
{
    var key:String;
    for (key in infoObject)
    {
        trace(key + ":" + infoObject[key]);
    }
}
```

显示以下输出：

```
parameters:  
name: point1  
time: 0.418  
type: navigation
```

此代码使用多种技术之一来设置运行回调方法的对象。您可以使用其它技术；有关详细信息，请参阅第 450 页的“[编写元数据和提示点的回调方法](#)”。

使用视频元数据

可以使用 `OnMetaData()` 和 `OnXMPData()` 函数来访问视频文件中的元数据信息，包括提示点。

使用 `OnMetaData()`

元数据包含有关视频文件的信息，如持续时间、宽度、高度和帧速率。添加到视频文件中的元数据信息取决于您用来编码视频文件的软件。

```
var nc:NetConnection = new NetConnection();  
nc.connect(null);  
  
var ns:NetStream = new NetStream(nc);  
ns.client = this;  
ns.play("video.flv");  
  
var vid:Video = new Video();  
vid.attachNetStream(ns);  
addChild(vid);  
  
function onMetaData(infoObject:Object):void  
{  
    var key:String;  
    for (key in infoObject)  
    {  
        trace(key + ": " + infoObject[key]);  
    }  
}
```

上面的代码生成如下输出：

```
width: 320  
audiodelay: 0.038  
canSeekToEnd: true  
height: 213  
cuePoints: ,  
audiodatarate: 96  
duration: 16.334  
videodatarate: 400  
framerate: 15  
videocodecid: 4  
audiocodecid: 2
```

 如果您的视频没有音频，则与音频相关的元数据信息（如 `audiodatarate`）将返回 `undefined`，因为在编码期间没有将音频信息添加到元数据中。

在上面的代码中，未显示提示点信息。为了显示提示点元数据，可以使用以下函数，该函数会以递归方式显示 `Object` 中的项目：

```
function traceObject(obj:Object, indent:uint = 0):void
{
    var indentString:String = "";
    var i:uint;
    var prop:String;
    var val.*;
    for (i = 0; i < indent; i++)
    {
        indentString += "\t";
    }
    for (prop in obj)
    {
        val = obj[prop];
        if (typeof(val) == "object")
        {
            trace(indentString + " " + prop + ": [Object]");
            traceObject(val, indent + 1);
        }
        else
        {
            trace(indentString + " " + prop + ": " + val);
        }
    }
}
```

使用上面的代码片断在 `onMetaData()` 方法中跟踪 `infoObject` 参数将产生以下输出：

```
width: 320
audiodatarate: 96
audiocodecid: 2
videocodecid: 4
videodatarate: 400
canSeekToEnd: true
duration: 16.334
audiodelay: 0.038
height: 213
framerate: 15
cuePoints: [Object]
  0: [Object]
    parameters: [Object]
      lights: beginning
      name: point1
      time: 0.418
      type: navigation
  1: [Object]
    parameters: [Object]
      lights: middle
      name: point2
      time: 7.748
      type: navigation
  2: [Object]
    parameters: [Object]
      lights: end
      name: point3
      time: 16.02
      type: navigation
```

下面的示例显示 MP4 视频的元数据。它假设存在一个名为 `metaDataOut` 的 `TextArea` 对象，它将把元数据写入其中。

```
package
{
    import flash.net.NetConnection;
    import flash.net.NetStream;
    import flash.events.NetStatusEvent;
    import flash.media.Video;
    import flash.display.StageDisplayState;
    import flash.display.Loader;
    import flash.display.Sprite;
    import flash.events.MouseEvent;

    public class onMetaDataExample extends Sprite
    {
        var video:Video = new Video();

        public function onMetaDataExample():void
        {
            var videoConnection:NetConnection = new NetConnection();
            videoConnection.connect(null);

            var videoStream:NetStream = new NetStream(videoConnection);
            videoStream.client = this;

            addChild(video);
            video.x = 185;
            video.y = 5;

            video.attachNetStream(videoStream);

            videoStream.play("video.mp4");

            videoStream.addEventListener(NetStatusEvent.NET_STATUS, netStatusHandler);
        }

        public function onMetaData(infoObject:Object):void
        {
            for(var propName:String in infoObject)
            {
                metaDataOut.appendText(propName + "=" + infoObject[propName] + "\n");
            }
        }

        private function netStatusHandler(event:NetStatusEvent):void
        {
            if(event.info.code == "NetStream.Play.Stop")
                stage.displayState = StageDisplayState.NORMAL;
        }
    }
}
```

对于此视频，onMetaData() 函数将生成以下输出：

```
moovposition=731965
height=352
avclevel=21
videocodecid=avc1
duration=2.36
width=704
videoframerate=25
avcprofile=88
trackinfo=[object Object]
```

使用信息对象

下表显示了在 `onMetaData()` 回调函数接收的 `Object` 中传递给 `onMetaData()` 回调函数的可能的视频元数据值。

参数	描述
<code>aacaot</code>	AAC 音频对象类型；支持 0、1 或 2。
<code>avclevel</code>	AVC IDC 级别号，如 10、11、20、21 等等。
<code>avcprofile</code>	AVC 配置文件号，如 55、77、100 等等。
<code>audiocodecid</code>	指示所用音频编解码器（编码 / 解码技术）的字符串 -- 例如“Mp3”或“mp4a”
<code>audiodatarate</code>	一个数字，指示音频的编码速率，以每秒千字节为单位。
<code>audiodelay</code>	一个数字，指示原始 FLV 文件的“time 0”在 FLV 文件中保持多长时间。为了正确同步音频，视频内容需要有少量的延迟。
<code>canSeekToEnd</code>	一个布尔值，如果 FLV 文件是用最后一帧（它允许定位到渐进式下载视频文件的末尾）上的关键帧编码的，则该值为 <code>true</code> 。如果 FLV 文件不是用最后一帧上的关键帧编码的，则该值为 <code>false</code> 。
<code>cuePoints</code>	嵌入在 FLV 文件中的提示点对象组成的数组，每个提示点对应一个对象。如果 FLV 文件不包含任何提示点，则值是未定义的。每个对象都具有以下属性：
	<ul style="list-style-type: none"> • <code>type</code>: 一个字符串，它将提示点的类型指定为“navigation”或“event”。 • <code>name</code>: 一个字符串，表示提示点的名称。 • <code>time</code>: 一个数字，表示以秒为单位的提示点的时间，精确到 3 位小数（毫秒）。 • <code>parameters</code>: 一个可选对象，包含用户在创建提示点时指定的名称 - 值对。
<code>duration</code>	一个数字，以秒为单位指定视频文件的持续时间。
<code>framerate</code>	一个数字，表示 FLV 文件的帧速率。
<code>height</code>	一个数字，以像素为单位表示 FLV 文件的高度。
<code>seekpoints</code>	一个数组，其中将可用的关键帧作为时间戳列出，单位为毫秒。可选。
<code>tags</code>	一个键 - 值对数组，这些键 / 值对表示“ilst”原子中的信息，相当于 MP4 文件中的 ID3 标记。iTunes 使用这些标记。可用于显示插图（如果有插图）。
<code>trackinfo</code>	一个对象，提供有关 MP4 文件中所有轨道的信息（包括其采样描述 ID）。
<code>videocodecid</code>	一个字符串，表示用于编码视频的编解码器版本。- 例如，“avc1”或“VP6F”
<code>videodatarate</code>	一个数字，表示 FLV 文件的视频数据速率。
<code>videoframerate</code>	MP4 视频的帧速率。
<code>width</code>	一个数字，以像素为单位表示 FLV 文件的宽度。

下表显示 `videocodecid` 参数的可能值：

<code>videocodecid</code>	编解码器名称
2	Sorenson H.263
3	屏幕视频（仅限 SWF 版本 7 和更高版本）
4	VP6（仅限 SWF 版本 8 和更高版本）
5	带有 Alpha 通道的 VP6 视频（仅限 SWF 版本 8 和更高版本）

下表显示 `audiocodecid` 参数的可能值：

audiocodecid	编解码器名称
0	未压缩
1	ADPCM
2	Mp3
4	Nellymoser, 16 kHz 单声
5	Nellymoser, 8kHz 单声
6	Nellymoser
10	AAC
11	Speex

使用 onXMPData()

onXMPData() 回调函数接收 Adobe F4V 或 FLV 视频文件中嵌入的 Adobe 可扩展元数据平台 (XMP) 专用信息。XMP 元数据包括提示点以及其它视频元数据。Flash Player 10 和 Adobe AIR 1.5 中引入了 XMP 元数据支持，并且 Flash Player 和 AIR 的后续版本都将支持这种元数据。

下面的示例处理 XMP 元数据中的提示点数据：

```
package
{
    import flash.display.*;
    import flash.net.*;
    import flash.events.NetStatusEvent;
    import flash.media.Video;

    public class onXMPDataExample extends Sprite
    {
        public function onXMPDataExample():void
        {
            var videoConnection:NetConnection = new NetConnection();
            videoConnection.connect(null);

            var videoStream:NetStream = new NetStream(videoConnection);
            videoStream.client = this;
            var video:Video = new Video();

            addChild(video);

            video.attachNetStream(videoStream);

            videoStream.play("video.f4v");
        }

        public function onMetaData(info:Object):void {
            trace("onMetaData fired");
        }

        public function onXMPData(infoObject:Object):void
        {
            trace("onXMPData Fired\n");
            //trace("raw XMP =\n");
            //trace(infoObject.data);
            var cuePoints:Array = new Array();
            var cuePoint:Object;
            var strFrameRate:String;
```

```
var nTracksFrameRate:Number;
var strTracks:String = "";
var onXMPXML = new XML(infoObject.data);
// Set up namespaces to make referencing easier
var xmpDM:Namespace = new Namespace("http://ns.adobe.com/xmp/1.0/DynamicMedia/");
var rdf:Namespace = new Namespace("http://www.w3.org/1999/02/22-rdf-syntax-ns#");
for each (var it:XML in onXMPXML..xmpDM::Tracks)
{
    var strTrackName:String = it.rdf::Bag.rdf::li.rdf::Description.@xmpDM::trackName;
    var strFrameRateXML:String = it.rdf::Bag.rdf::li.rdf::Description.@xmpDM::frameRate;
    strFrameRate = strFrameRateXML.substr(1,strFrameRateXML.length);

    nTracksFrameRate = Number(strFrameRate);

    strTracks += it;
}
var onXMPTracksXML:XML = new XML(strTracks);
var strCuepoints:String = "";
for each (var item:XML in onXMPTracksXML..xmpDM::markers)
{
    strCuepoints += item;
}
trace(strCuepoints);
}
```

对于名为 startrekintro.f4v 的短视频文件，此示例生成以下跟踪行。这些行显示了 XMP 元数据中用于导航的提示点数据以及事件提示点：

```
onMetaData fired
onXMPData Fired

<xmpDM:markers xmlns:xmp="http://ns.adobe.com/xap/1.0/"
  xmlns:xmpDM="http://ns.adobe.com/xmp/1.0/DynamicMedia/"
  xmlns:stDim="http://ns.adobe.com/xap/1.0/sType/Dimensions#" xmlns:xmpMM="http://ns.adobe.com/xap/1.0/mm/"
  xmlns:stEvt="http://ns.adobe.com/xap/1.0/sType/ResourceEvent#"
  xmlns:dc="http://purl.org/dc/elements/1.1/" xmlns:rdf="http://www.w3.org/1999/02/22-rdf-syntax-ns#"
  xmlns:x="adobe:ns:meta/">
  <rdf:Seq>
    <rdf:li>
      <rdf:Description xmpDM:startTime="7695905817600" xmpDM:name="Title1" xmpDM:type="FLVCuePoint"
        xmpDM:cuePointType="Navigation">
        <xmpDM:cuePointParams>
          <rdf:Seq>
            <rdf:li xmpDM:key="Title" xmpDM:value="Star Trek"/>
            <rdf:li xmpDM:key="Color" xmpDM:value="Blue"/>
          </rdf:Seq>
        </xmpDM:cuePointParams>
      </rdf:Description>
    </rdf:li>
    <rdf:li>
      <rdf:Description xmpDM:startTime="10289459980800" xmpDM:name="Title2" xmpDM:type="FLVCuePoint"
        xmpDM:cuePointType="Event">
        <xmpDM:cuePointParams>
          <rdf:Seq>
            <rdf:li xmpDM:key="William Shatner" xmpDM:value="First Star"/>
            <rdf:li xmpDM:key="Color" xmpDM:value="Light Blue"/>
          </rdf:Seq>
        </xmpDM:cuePointParams>
      </rdf:Description>
    </rdf:li>
  </rdf:Seq>
</xmpDM:markers>
onMetaData fired
```

注：在 XMP 数据中，时间是作为 DVA 刻度存储的，而不是秒。若要计算提示点时间，请将启动时间除以帧速率。例如，启动时间 7695905817600 除以帧速率 254016000000 等于 30:30。

若要查看全部原始 XMP 元数据，包括帧速率，请删除位于 onXMPData() 函数开头的第二和第三条 trace() 语句之前的注释标识符 (//)。

有关 XMP 的详细信息，请参阅：

- <http://partners.adobe.com/public/developer/xmp/topic.html>
- <http://www.adobe.com/devnet/xmp/>

使用图像元数据

onImageData 事件通过 AMFO 数据通道将图像数据作为字节数组发送。该数据可以是 JPEG、PNG 或 GIF 格式。可定义 onImageData() 回调方法来处理此信息，这与定义 onCuePoint 和 onMetaData 的回调方法的方式相同。下面的示例使用 onImageData() 回调方法访问并显示图像数据。

```

public function onImageData(imageData:Object):void
{
    // display track number
    trace(imageData.trackid);
    var loader:Loader = new Loader();
    //imageData.data is a ByteArray object
    loader.loadBytes(imageData.data);
    addChild(loader);
}

```

使用文本元数据

`onTextData` 事件通过 AMFO 数据通道发送文本数据。文本数据为 **UTF-8** 格式，并且包含有关基于 **3GP** 计时文本规范的格式化的更多信息。此规范定义了标准化的字幕格式。可定义 `onTextData()` 回调方法来处理此信息，这与定义 `onCuePoint` 或 `onMetaData` 的回调方法的方式相同。在下面的示例中，`onTextData()` 方法显示曲目 ID 号和对应的曲目文本。

```

public function onTextData(textData:Object):void
{
    // display the track number
    trace(textData.trackid);
    // displays the text, which can be a null string, indicating old text
    // that should be erased
    trace(textData.text);
}

```

捕获摄像头输入

除了外部视频文件外，附加到用户计算机上的摄像头也可以作为您使用 ActionScript 进行显示和操作的视频数据的来源。`Camera` 类是 ActionScript 中内置的机制，用于使用计算机摄像头。

了解 Camera 类

使用 `Camera` 对象可以连接到用户的本地摄像头并在本地广播视频（播放给用户），或将其广播到远程服务器（比如 Flash Media Server）。

使用 `Camera` 类可以访问有关用户摄像头的以下各种信息：

- Flash Player 可以使用用户计算机上安装的哪些摄像头
- 是否安装了摄像头
- 允许还是拒绝 Flash Player 访问用户的摄像头
- 哪个摄像头当前处于活动状态
- 正在捕获的视频的宽度和高度

`Camera` 类包括多个有用的方法和属性，通过这些方法和属性可以使用 `Camera` 对象。例如，静态 `Camera.names` 属性包含由当前安装在用户计算机上的摄像头的名称所组成的数组。您也可以使用 `name` 属性显示当前处于活动状态的摄像头的名称。

在屏幕上显示摄像头内容

连接到摄像头所需的代码比使用 `NetConnection` 和 `NetStream` 类加载视频的代码要少。由于在使用 Flash Player 时，您需要先请求用户允许您连接其摄像头，然后才能访问这些摄像头，因此，`Camera` 类可能很可能变得非常麻烦。

以下代码演示如何使用 `Camera` 类连接到用户的本地摄像头：

```
var cam:Camera = Camera.getCamera();
var vid:Video = new Video();
vid.attachCamera(cam);
addChild(vid);
```

注: Camera 类没有构造函数方法。若要创建新的 Camera 实例, 请使用静态 Camera.getCamera() 方法。

设计摄像头应用程序

在编写需要连接到用户摄像头的应用程序时, 需要在代码中考虑以下事项:

- 检查用户当前是否安装了摄像头。
- (只针对 Flash Player) 检查用户是否显式允许访问其摄像头。出于安全原因, 播放器会显示“Flash Player 设置”对话框, 让用户选择允许还是拒绝对其摄像头的访问。这样可以防止 Flash Player 在未经用户许可的情况下连接到其摄像头并广播视频流。如果用户单击允许, 则应用程序即可连接到用户的摄像头。如果用户单击拒绝, 则应用程序将无法访问用户的摄像头。应用程序始终应适当地处理这两种情况。

连接至用户的摄像头

连接到用户摄像头时, 执行的第一步是通过创建一个类型为 Camera 的变量并将其初始化为静态 Camera.getCamera() 方法的返回值来创建一个新的 Camera 实例。

下一步是创建一个新的视频对象并向其附加 Camera 对象。

第三步是向显示列表中添加该视频对象。由于 Camera 类不会扩展 DisplayObject 类, 它不能直接添加到显示列表中, 因此需要执行第 2 步和第 3 步。若要显示摄像头捕获的视频, 需要创建一个新的视频对象并调用 attachCamera() 方法。

以下代码演示这三个步骤:

```
var cam:Camera = Camera.getCamera();
var vid:Video = new Video();
vid.attachCamera(cam);
addChild(vid);
```

注意, 如果用户未安装摄像头, 应用程序不会显示任何内容。

在实际情况下, 您需要对应用程序执行其它步骤。有关详细信息, 请参阅第 463 页的“[验证是否已安装摄像头](#)”和第 464 页的“[检测摄像头的访问权限](#)”。

验证是否已安装摄像头

在尝试对 Camera 实例使用任何方法或属性之前, 您需要验证用户是否已安装了摄像头。检查用户是否已安装摄像头有两种方式:

- 检查静态 Camera.names 属性, 该属性包含可用摄像头名称的数组。此数组通常具有一个或几个字符串, 因为多数用户不太可能同时安装多个摄像头。以下代码演示如何检查 Camera.names 属性以查看用户是否具有可用的摄像头:

```
if (Camera.names.length > 0)
{
    trace("User has at least one camera installed.");
    var cam:Camera = Camera.getCamera(); // Get default camera.
}
else
{
    trace("User has no cameras installed.");
```

- 检查静态 Camera.getCamera() 方法的返回值。如果没有摄像头可用或未安装摄像头, 则此方法将返回 null, 否则返回对 Camera 对象的引用。以下代码演示如何检查 Camera.getCamera() 方法以查看用户是否具有可用的摄像头:

```

var cam:Camera = Camera.getCamera();
if (cam == null)
{
    trace("User has no cameras installed.");
}
else
{
    trace("User has at least 1 camera installed.");
}

```

由于 Camera 类未扩展 DisplayObject 类，因此无法使用 addChild() 方法将其直接添加到显示列表。为了显示摄像头捕获的视频，您需要创建一个新的 Video 对象并调用该 Video 实例的 attachCamera() 方法。

以下代码片断演示在存在摄像头的情况下如何附加摄像头；如果不存在摄像头，应用程序将不显示任何内容：

```

var cam:Camera = Camera.getCamera();
if (cam != null)
{
    var vid:Video = new Video();
    vid.attachCamera(cam);
    addChild(vid);
}

```

检测摄像头的访问权限

在 AIR 应用程序沙箱中，应用程序无需用户许可即可访问任何摄像头。

在 Flash Player 可显示摄像头输出之前，用户必须显式允许 Flash Player 访问摄像头。在调用 attachCamera() 方法后，Flash Player 会显示“Flash Player 设置”对话框，提示用户允许或拒绝 Flash Player 访问摄像头和麦克风。如果用户单击“允许”按钮，Flash Player 将在舞台上的 Video 实例中显示摄像头输出。如果用户单击“拒绝”按钮，则 Flash Player 将无法连接到摄像头，且 Video 对象将不显示任何内容。

如果想要检测用户是否允许 Flash Player 访问其摄像头，可以侦听摄像头的 status 事件 (StatusEvent.STATUS)，如以下代码所示：

```

var cam:Camera = Camera.getCamera();
if (cam != null)
{
    cam.addEventListener(StatusEvent.STATUS, statusHandler);
    var vid:Video = new Video();
    vid.attachCamera(cam);
    addChild(vid);
}
function statusHandler(event:StatusEvent):void
{
    // This event gets dispatched when the user clicks the "Allow" or "Deny"
    // button in the Flash Player Settings dialog box.
    trace(event.code); // "Camera.Muted" or "Camera.Unmuted"
}

```

一旦用户单击“允许”或“拒绝”，即会调用 statusHandler() 函数。使用以下两种方法之一可以检测用户单击了哪个按钮：

- statusHandler() 函数的 event 参数包含一个 code 属性，其中含有字符串“Camera.Muted”或“Camera.Unmuted”。如果该值为“Camera.Muted”，表示用户单击了“拒绝”按钮，因此 Flash Player 无法访问摄像头。在下面的代码片断中您会看到此情况的一个示例：

```

function statusHandler(event:StatusEvent):void
{
    switch (event.code)
    {
        case "Camera.Muted":
            trace("User clicked Deny.");
            break;
        case "Camera.Unmuted":
            trace("User clicked Accept.");
            break;
    }
}

```

- Camera 类包含一个名为 muted 的只读属性，它指明用户在 Flash Player“隐私”面板中是拒绝访问摄像头 (true) 还是允许访问摄像头 (false)。在下面的代码片断中您会看到此情况的一个示例：

```

function statusHandler(event:StatusEvent):void
{
    if (cam.muted)
    {
        trace("User clicked Deny.");
    }
    else
    {
        trace("User clicked Accept.");
    }
}

```

通过检查将要调度的 status 事件，您可以编写代码来处理用户接受或拒绝访问摄像头的操作并进行相应的清理。例如，如果用户单击“拒绝”按钮，则您可以向用户显示一条消息，说明他们如果想要参加视频聊天的话，需要单击“允许”；或者，您也可以确保删除显示列表中的 Video 对象以释放系统资源。

最优化视频品质

默认情况下，Video 类的新实例为 320 像素宽乘以 240 像素高。为了最优化视频品质，应始终确保视频对象与 Camera 对象返回的视频具有相同的尺寸。使用 Camera 类的 width 和 height 属性，您可以获取 Camera 对象的宽度和高度，然后将该视频对象的 width 和 height 属性设置为与 Camera 对象的尺寸相符，也可以将 Camera 对象的宽度和高度传递给 Video 类的构造函数方法，如以下代码片断所示：

```

var cam:Camera = Camera.getCamera();
if (cam != null)
{
    var vid:Video = new Video(cam.width, cam.height);
    vid.attachCamera(cam);
    addChild(vid);
}

```

由于 getCamera() 方法返回对 Camera 对象的引用（在没有可用摄像头时返回 null），因此，即使用户拒绝访问其摄像头，您也可以访问 Camera 对象的方法和属性。这样可以使用摄像头的本机高度和宽度设置视频实例的尺寸。

```
var vid:Video;
var cam:Camera = Camera.getCamera();

if (cam == null)
{
    trace("Unable to locate available cameras.");
}
else
{
    trace("Found camera: " + cam.name);
    cam.addEventListener(StatusEvent.STATUS, statusHandler);
    vid = new Video();
    vid.attachCamera(cam);
}

function statusHandler(event:StatusEvent):void
{
    if (cam.muted)
    {
        trace("Unable to connect to active camera.");
    }
    else
    {
        // Resize Video object to match camera settings and
        // add the video to the display list.
        vid.width = cam.width;
        vid.height = cam.height;
        addChild(vid);
    }
    // Remove the status event listener.
    cam.removeEventListener(StatusEvent.STATUS, statusHandler);
}
```

有关全屏模式的信息，请参阅第 236 页的“[设置舞台属性](#)”下的全屏模式部分。

监视播放条件

Camera 类包含多个属性，这些属性允许您监视 Camera 对象的当前状态。例如，以下代码使用一个 Timer 对象和一个文本字段实例在显示列表上显示摄像头的若干属性：

```
var vid:Video;
var cam:Camera = Camera.getCamera();
var tf:TextField = new TextField();
tf.x = 300;
tf.autoSize = TextFieldAutoSize.LEFT;
addChild(tf);

if (cam != null)
{
    cam.addEventListener(StatusEvent.STATUS, statusHandler);
    vid = new Video();
    vid.attachCamera(cam);
}

function statusHandler(event:StatusEvent):void
{
    if (!cam.muted)
    {
        vid.width = cam.width;
        vid.height = cam.height;
        addChild(vid);
```

```
t.start();
}
cam.removeEventListener(StatusEvent.STATUS, statusHandler);
}

var t:Timer = new Timer(100);
t.addEventListener(TimerEvent.TIMER, timerHandler);
function timerHandler(event:TimerEvent):void
{
    tf.text = "";
    tf.appendText("activityLevel: " + cam.activityLevel + "\n");
    tf.appendText("bandwidth: " + cam.bandwidth + "\n");
    tf.appendText("currentFPS: " + cam.currentFPS + "\n");
    tf.appendText("fps: " + cam.fps + "\n");
    tf.appendText("keyFrameInterval: " + cam.keyFrameInterval + "\n");
    tf.appendText("loopback: " + cam.loopback + "\n");
    tf.appendText("motionLevel: " + cam.motionLevel + "\n");
    tf.appendText("motionTimeout: " + cam.motionTimeout + "\n");
    tf.appendText("quality: " + cam.quality + "\n");
}
```

每 1/10 秒（100 毫秒），即会调度一次 Timer 对象的 timer 事件，并且 timerHandler() 函数会更新显示列表上的文本字段。

向服务器发送视频

如果您要生成涉及 Video 或 Camera 对象的更为复杂的应用程序，可以使用 Flash Media Server 提供的流媒体功能和开发环境组合来创建媒体应用程序并将它提供给广泛的目标用户。开发人员可以使用这一组合来创建应用程序，如视频点播、实时 Web 事件广播和 Mp3 流式传输，以及视频博客、视频消息传送和多媒体聊天环境。有关详细信息，请参阅 www.adobe.com/go/learn_fms_docs_cn 上提供的在线 Flash Media Server 文档。

FLV 文件的高级主题

以下主题介绍使用 FLV 文件的一些特殊问题。

关于配置 FLV 文件以便在服务器上托管

在处理 FLV 文件时，您可能需要配置服务器以便处理 FLV 文件格式。多用途 Internet 邮件扩展 (MIME) 是标准的数据规范，允许您通过 Internet 连接发送非 ASCII 文件。Web 浏览器和电子邮件客户端经过配置，可以解释多种 MIME 类型，因此它们可以发送和接收视频、音频、图形和格式化文本。若要从 Web 服务器加载 FLV 文件，则可能需要向您的 Web 服务器注册文件扩展名和 MIME 类型，因此应当检查您的 Web 服务器文档。FLV 文件的 MIME 类型是 video/x-flv。下面列出了 FLV 文件类型的完整信息：

- Mime 类型: video/x-flv
- 文件扩展名: .flv
- 必需参数: 无
- 可选参数: 无
- 编码注意事项: FLV 文件是二进制文件；有些应用程序可能需要设置应用程序 / 八位字节流子类型
- 安全问题: 无
- 已发布的规范: www.adobe.com/go/video_file_format_cn

Microsoft 改变了在 Microsoft Internet 信息服务 (IIS) 6.0 Web 服务器中处理流媒体的方式，不再采用早期版本中的处理方式。早期版本的 IIS 不需要对 Flash 视频流做任何修改。在 Windows 2003 附带的默认 Web 服务器 IIS 6.0 中，服务器需要借助 MIME 类型来确认 FLV 文件为流媒体。

当采用流式媒体的方式加载外部 FLV 文件的 SWF 文件被置于 Microsoft Windows Server® 2003 上，并在浏览器中查看时，可以正确播放 SWF 文件，但 FLV 视频却不能采用流式媒体的方式加载。这个问题会影响到放置在 Windows Server 2003 上的所有 FLV 文件，包括用早期版本的 Flash 创作工具以及 Adobe 的 Macromedia Flash Video Kit for Dreamweaver MX 2004 制作的文件。如果在其它操作系统上对这些文件进行测试，则这些文件可以正常工作。

有关配置 Microsoft Windows 2003 和 Microsoft IIS Server 6.0 以流式传输 FLV 视频的信息，请参阅 www.adobe.com/go/tn_19439_cn。

关于在 Macintosh 上将本地 FLV 文件设定为目标

如果您尝试从 Apple® Macintosh® 计算机的非系统驱动器上，使用以斜杠 (/) 表示的相对路径来播放本地 FLV，将无法播放视频。非系统驱动器包括（但不限于）CD-ROM、分区的硬盘、可移除的存储媒体以及连接的存储设备。

注：导致失败发生的原因是操作系统的局限，而非 Flash Player 或 AIR 的局限。

若要从 Macintosh 的非系统驱动器上播放 FLV 文件，需使用基于冒号记号 (:) 的绝对路径来代替基于斜杠记号 (/) 的路径引用该文件。下面的列表显示了这两种记号的不同：

- 基于斜杠的记号：myDrive/myFolder/myFLV.flv
- 基于冒号的记号：(Mac OS*) myDrive:myFolder:myFLV.flv

您还可以为想让 Macintosh 播放的 CD-ROM 创建一个放映文件。有关 Mac OS CD-ROM 和 FLV 文件的最新信息，请访问 www.adobe.com/go/3121b301_cn。

示例：视频自动唱片点唱机

以下示例生成一个简单的视频自动唱片点唱机，它可以动态加载一组视频并按顺序播放。您可以生成一个可以让用户浏览一系列视频教程的应用程序，也可以指定在传送用户请求的视频之前应播放的广告。此示例演示了 ActionScript 3.0 的下列功能：

- 根据视频文件的播放进度更新播放头
- 监听并分析视频文件的元数据
- 处理网络流中的特定代码
- 加载、播放、暂停和停止动态加载的 FLV
- 根据网络流的元数据调整显示列表上视频对象的大小

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。“视频自动唱片点唱机”应用程序文件位于 Samples/VideoJukebox 文件夹中。该应用程序包含以下文件：

文件	描述
VideoJukebox.fla 或 VideoJukebox.mxml	Flex 或 Flash 的主应用程序文件（分别为 MXML 和 FLA 格式）。
VideoJukebox.as	此类提供了应用程序的主要功能。
playlist.xml	列出要向视频自动唱片点唱机中加载哪些视频文件的文件。

加载外部视频播放列表文件

外部 `playlist.xml` 文件指定要加载的视频以及播放这些文件的顺序。为了加载该 XML 文件，您需要使用一个 `URLLoader` 对象和一个 `URLRequest` 对象，如以下代码所示：

```
uldr = new URLLoader();
uldr.addEventListener(Event.COMPLETE, xmlCompleteHandler);
uldr.load(new URLRequest(PLAYLIST_XML_URL));
```

此代码放置在 `VideoJukebox` 类的构造函数内，因此，在运行其它任何代码之前，会首先加载该文件。一旦 XML 文件加载完成后，即会调用 `xmlCompleteHandler()` 方法，该方法会将该外部文件分析为一个 XML 对象，如以下代码所示：

```
private function xmlCompleteHandler(event:Event):void
{
    playlist = XML(event.target.data);
    videosXML = playlist.video;
    main();
}
```

`playlist` XML 对象包含来自该外部文件的原始 XML，而 `videosXML` 是仅包含视频节点的 `XMLElement` 对象。以下代码片断显示了一个示例 `playlist.xml` 文件：

```
<videos>
    <video url="video/caption_video.flv" />
    <video url="video/cuepoints.flv" />
    <video url="video/water.flv" />
</videos>
```

最后，`xmlCompleteHandler()` 方法将调用 `main()` 方法，被调用的方法会在显示列表上设置各个组件实例以及用于加载外部 FLV 文件的 `NetConnection` 和 `NetStream` 对象。

创建用户界面

若要构建用户界面，您需要将 5 个 `Button` 实例拖到显示列表上并为其指定以下实例名称：`playButton`、`pauseButton`、`stopButton`、`backButton` 和 `forwardButton`。

对于这些 `Button` 实例中的每个实例，您需要为 `click` 事件分配一个处理函数，如以下代码片断所示：

```
playButton.addEventListener(MouseEvent.CLICK, buttonClickHandler);
pauseButton.addEventListener(MouseEvent.CLICK, buttonClickHandler);
stopButton.addEventListener(MouseEvent.CLICK, buttonClickHandler);
backButton.addEventListener(MouseEvent.CLICK, buttonClickHandler);
forwardButton.addEventListener(MouseEvent.CLICK, buttonClickHandler);
```

`buttonClickHandler()` 方法使用 `switch` 语句来确定单击了哪个按钮实例，如以下代码所示：

```

private function buttonClickHandler(event:MouseEvent):void
{
    switch (event.currentTarget)
    {
        case playButton:
            ns.resume();
            break;
        case pauseButton:
            ns.togglePause();
            break;
        case stopButton:
            ns.pause();
            ns.seek(0);
            break;
        case backButton:
            playPreviousVideo();
            break;
        case forwardButton:
            playNextVideo();
            break;
    }
}

```

接下来，向显示列表中添加一个 **Slider** 实例，并为其指定实例名称 **volumeSlider**。以下代码将该滑块实例的 **liveDragging** 属性设置为 **true**，并为该滑块实例的 **change** 事件定义一个事件侦听器：

```

volumeSlider.value = volumeTransform.volume;
volumeSlider.minimum = 0;
volumeSlider.maximum = 1;
volumeSlider.snapInterval = 0.1;
volumeSlider.tickInterval = volumeSlider.snapInterval;
volumeSlider.liveDragging = true;
volumeSlider.addEventListener(SliderEvent.CHANGE, volumeChangeHandler);

```

向显示列表中添加一个 **ProgressBar** 实例，并为其指定实例名称 **positionBar**。将其 **mode** 属性设置为 **manual**，如以下代码片断所示：

```
positionBar.mode = ProgressBarMode.MANUAL;
```

最后，向显示列表中添加一个 **Label** 实例，并为其指定实例名称 **positionLabel**。此 **Label** 实例的值将由计时器实例设置

侦听视频对象的元数据

当 Flash Player 遇到已加载的每个视频的元数据时，会对 **NetStream** 对象的 **client** 属性调用 **onMetaData()** 回调处理函数。以下代码初始化一个对象并设置指定的回调处理函数：

```

client = new Object();
client.onMetaData = metadataHandler;

```

metadataHandler() 方法将其数据复制到在代码前面部分中定义的 **meta** 属性中。这可使您在应用程序运行过程中随时访问当前视频的元数据。接下来，调整舞台上视频对象的大小，使其与从元数据中返回的尺寸相符。最后，根据当前正在播放的视频的尺寸移动并调整 **positionBar** 进度栏实例的大小。以下代码包含完整的 **metadataHandler()** 方法：

```

private function metadataHandler(metadataObj:Object):void
{
    meta = metadataObj;
    vid.width = meta.width;
    vid.height = meta.height;
    positionBar.move(vid.x, vid.y + vid.height);
    positionBar.width = vid.width;
}

```

动态加载 Flash 视频

为了动态加载每个 Flash 视频，应用程序应使用 NetConnection 和 NetStream 对象。以下代码创建一个 NetConnection 对象并将 null 传递给 connect() 方法。通过指定 null，Flash Player 会连接到本地服务器上的视频，而不连接到服务器（如 Flash Media Server）。

以下代码创建 NetConnection 和 NetStream 实例，为 netStatus 事件定义事件侦听器，并将 client 对象分配给 client 属性：

```
nc = new NetConnection();
nc.connect(null);

ns = new NetStream(nc);
ns.addEventListener(NetStatusEvent.NET_STATUS, netStatusHandler);
ns.client = client;
```

每当视频的状态发生改变时即会调用 netStatusHandler() 方法。这包括视频开始或停止播放时、缓冲时或找不到视频流时。以下代码列出了 netStatusHandler() 事件：

```
private function netStatusHandler(event:NetStatusEvent):void
{
    try
    {
        switch (event.info.code)
        {
            case "NetStream.Play.Start":
                t.start();
                break;
            case "NetStream.Play.StreamNotFound":
            case "NetStream.Play.Stop":
                t.stop();
                playNextVideo();
                break;
        }
    }
    catch (error:TypeError)
    {
        // Ignore any errors.
    }
}
```

上面的代码计算 info 对象的 code 属性，并过滤出“NetStream.Play.Start”、“NetStream.Play.StreamNotFound”或“NetStream.Play.Stop”的代码。其它所有代码均被忽略。如果网络流已开始，代码会启动用于更新播放头的 Timer 实例。如果找不到网络流或网络流已停止，则会停止 Timer 实例，应用程序将尝试播放播放列表中的下一个视频。

每次执行 Timer 时，positionBar 进度栏实例都会通过调用 ProgressBar 类的 setProgress() 方法来更新其当前位置，并且会用当前视频已经过的时间和总时间来更新 positionLabel Label 实例。

```
private function timerHandler(event:TimerEvent):void
{
    try
    {
        positionBar.setProgress(ns.time, meta.duration);
        positionLabel.text = ns.time.toFixed(1) + " of " + meta.duration.toFixed(1) + " seconds";
    }
    catch (error:Error)
    {
        // Ignore this error.
    }
}
```

控制视频音量

通过在 NetStream 对象上设置 soundTransform 属性，您可以控制动态加载的视频的音量。视频自动唱片点唱机应用程序允许您通过更改 volumeSlider Slider 实例的值来修改音量级别。以下代码演示如何通过将 Slider 组件的值赋给 SoundTransform 对象（在 NetStream 对象上设置为 soundTransform 属性）来更改音量级别：

```
private function volumeChangeHandler(event:SliderEvent):void
{
    volumeTransform.volume = event.value;
    ns.soundTransform = volumeTransform;
}
```

控制视频播放

应用程序的其余部分是在视频到达视频流末尾或用户跳到上一个或下一个视频时控制视频的播放。

下面的方法可以从当前所选索引的 XMLList 中检索视频 URL：

```
private function getVideo():String
{
    return videosXML[idx].@url;
}
```

playVideo() 方法会调用 NetStream 对象上的 play() 方法以加载当前所选的视频：

```
private function playVideo():void
{
    var url:String = getVideo();
    ns.play(url);
}
```

playPreviousVideo() 方法使当前视频索引递减、调用 playVideo() 方法来加载新的视频文件并将进度栏设置为可见：

```
private function playPreviousVideo():void
{
    if (idx > 0)
    {
        idx--;
        playVideo();
        positionBar.visible = true;
    }
}
```

最后一个方法 playNextVideo() 使视频索引递增并调用 playVideo() 方法。如果当前视频是播放列表中的最后一个视频，则会对 Video 对象调用 clear() 方法，并将进度栏实例的 visible 属性设置为 false：

```
private function playNextVideo():void
{
    if (idx < (videosXML.length() - 1))
    {
        idx++;
        playVideo();
        positionBar.visible = true;
    }
    else
    {
        idx++;
        vid.clear();
        positionBar.visible = false;
    }
}
```

第 25 章：处理声音

ActionScript 是为开发引人入胜的交互式应用程序而设计的，这种极其引人入胜的应用程序中经常被忽略的一个元素是声音。您可以在视频游戏中添加声音效果，在应用程序用户界面中添加音频回馈，甚至创建一个分析通过 Internet 加载的 mp3 文件的程序（在这些情况下，声音是应用程序的核心）。

在本章中，您将了解如何加载外部音频文件以及处理 SWF 中嵌入的音频。您还会了解如何控制音频、创建声音信息的可视表示形式以及从用户麦克风捕获声音。

声音处理基础知识

处理声音简介

计算机可以捕获和编码数字音频（声音信息的计算机表示形式），还可以存储和检索声音，从而通过扬声器播放。您可以使用 Adobe® Flash® Player 或 Adobe® AIR™ 以及 ActionScript 来播放声音。

将声音数据转换为数字形式后，它具有各种不同的特性，如声音的音量以及它是立体声还是单声道声音。在 ActionScript 中播放声音时，您也可以调整这些特性；例如，使声音变得更大，或者使其像是来自某个方向。

您需要先将声音信息加载到 Flash Player 或 AIR 中，才能在 ActionScript 中控制声音。可以使用五种方法将音频数据加载到 Flash Player 或 AIR 中，以便通过 ActionScript 处理这些数据。您可以将外部声音文件（如 mp3 文件）加载到 SWF 中；可以在创建 SWF 文件时将声音信息直接嵌入到其中；可以使用连接到用户计算机的麦克风来获取音频输入；可以访问从服务器流式传输的声音数据；还可以处理动态生成的声音数据。

从外部声音文件加载声音数据时，您可以在加载其余声音数据的同时开始播放声音文件的开头部分。

虽然可以使用各种不同的声音文件格式对数字音频进行编码，但是 ActionScript 3.0、Flash Player 和 AIR 支持以 mp3 格式存储的声音文件。它们不能直接加载或播放其它格式的声音文件，如 WAV 或 AIFF。

在 ActionScript 中处理声音时，可能会使用 flash.media 包中的某些类。Sound 类用于访问音频信息，方法是：加载声音文件或为对声音数据进行采样的事件分配函数，然后开始播放。开始播放声音后，Flash Player 和 AIR 为您提供对 SoundChannel 对象的访问。因为已加载的音频文件只能是您在用户计算机上播放的几种声音之一，所以，所播放的每种单独的声音都使用自己的 SoundChannel 对象；混合在一起的所有 SoundChannel 对象的组合输出是实际通过计算机扬声器播放的声音。可以使用此 SoundChannel 实例来控制声音的属性以及停止其播放。最后，如果要控制组合音频，您可以通过 SoundMixer 类对混合输出进行控制。

也可以使用其它几个类，在 ActionScript 中处理声音时执行更具体的任务；要了解与声音有关的所有类的详细信息，请参阅第 475 页的“[了解声音体系结构](#)”。

常见的声音处理任务

本章介绍以下您可能希望执行的与声音有关的任务：

- 加载外部 mp3 文件并跟踪其加载进度
- 播放、暂停、继续播放以及停止播放声音
- 在加载的同时播放声音流
- 控制音量和声相
- 从 mp3 文件中检索 ID3 元数据
- 使用原始声音波形数据

- 动态生成声音
- 从用户麦克风捕获并重新播放声音输入

重要概念和术语

以下参考列表包含将会在本章中遇到的重要术语：

- 波幅：声音波形上的点与零或平衡线之间的距离。
- 比特率：每秒为声音文件编码或流式传输的数据量。对于 mp3 文件，比特率通常是以每秒千位数 (kbps) 来表述的。较高的比特率通常意味着较高品质的声音波形。
- 缓冲：在播放之前接收和存储声音数据。
- mp3：MPEG-1 Audio Layer 3 (mp3) 是一种常用的声音压缩格式。
- 声相：将音频信号放在立体声声场中左声道和右声道之间。
- 峰值：波形中的最高点。
- 采样率：定义在生成数字信号时每秒从模拟音频信号采集的样本数。标准光盘音频的采样率为 44.1 kHz 或每秒 44,100 个样本。
- 流式传输：此过程是指，在从服务器加载声音文件或视频文件的后面部分的同时播放该文件的前面部分。
- 音量：声音的响度。
- 波形：声音信号波幅随时间变化的图形形状。

完成本章中的示例

在学习本章的过程中，您可能希望测试一些示例代码清单。因为本章介绍如何在 ActionScript 中处理声音，所以许多示例都涉及对声音文件的处理，例如，播放声音文件、停止播放或以某种方式调整声音。测试本章中的示例：

- 1 创建一个新的 Flash 文档并将它保存到您的计算机上。
- 2 在时间轴上，选择第一个关键帧，然后打开“动作”面板。
- 3 将示例代码清单复制到“脚本”窗格中。
- 4 如果代码涉及加载外部声音文件，则其中会有一行与下面的代码类似的代码：

```
var req:URLRequest = new URLRequest("click.mp3");
var s:Sound = new Sound(req);
s.play();
```

其中“click.mp3”是要加载的声音文件的名称。为了测试这些示例，需要有一个可使用的 mp3 文件。应将此 mp3 文件放入 Flash 文档所在的文件夹中。然后，应更改代码以使用您的 mp3 文件的名称，而不是代码清单中的名称（例如，在上面的代码中，将“click.mp3”改为您的 mp3 文件的名称）。

- 5 从主菜单中，选择“控制”>“测试影片”以创建 SWF 文件并预览（并且听一下）示例的输出。

除了播放音频外，某些示例使用 trace() 函数来显示值；当测试这些示例时，将在“输出”面板中看到这些值的结果。一些示例还将内容绘制到屏幕上，因此对于这些示例，您还将在 Flash Player 或 AIR 窗口中看到内容。

有关测试本手册中的示例代码清单的详细信息，请参阅第 29 页的“[测试本章内的示例代码清单](#)”。

了解声音体系结构

应用程序可以从以下五种主要来源加载声音数据：

- 在运行时加载的外部声音文件
- 在应用程序的 SWF 文件中嵌入的声音资源
- 来自连接到用户系统上的麦克风的声音数据
- 从远程媒体服务器流式传输的声音数据，如 Flash Media Server
- 通过使用 sampleData 事件处理函数动态生成的声音数据

可以在播放之前加载全部声音数据，也可以进行流式传输，即在进行加载的同时播放这些数据。

ActionScript 3.0 声音类支持以 mp3 格式存储的声音文件。它们不能直接加载或播放其它格式（如 WAV 或 AIFF）的声音文件。但从 Flash Player 9.0.115.0 开始，可使用 NetStream 类加载和播放 AAC 音频文件。这与加载和播放视频内容的方法相同。有关此方法的详细信息，请参阅第 440 页的“[处理视频](#)”。

使用 Adobe Flash CS4 Professional，可以导入 WAV 或 AIFF 声音文件，然后将其以 mp3 格式嵌入应用程序的 SWF 文件中。Flash 创作工具还可压缩嵌入的声音文件以减小文件大小，但会降低声音的品质。有关详细信息，请参阅《使用 Flash》中的“[导入声音](#)”。

ActionScript 3.0 声音体系结构使用 flash.media 包中的以下类。

类	描述
flash.media.Sound	Sound 类处理声音加载、管理基本声音属性以及启动声音播放。
flash.media.SoundChannel	当应用程序播放 Sound 对象时，将创建一个新的 SoundChannel 对象来控制回放。SoundChannel 对象控制声音的左和右回放声道的音量。播放的每种声音具有其自己的 SoundChannel 对象。
flash.media.SoundLoaderContext	SoundLoaderContext 类指定在加载声音时使用的缓冲秒数，以及 Flash Player 或 AIR 在加载文件时是否从服务器中查找策略文件。SoundLoaderContext 对象用作 Sound.load() 方法的参数。
flash.media.SoundMixer	SoundMixer 类控制与应用程序中的所有声音有关的回放和安全属性。实际上，可通过一个通用 SoundMixer 对象将多个声道混合在一起，因此，该 SoundMixer 对象中的属性值将影响当前播放的所有 SoundChannel 对象。
flash.media.SoundTransform	SoundTransform 类包含控制音量和声相的值。可以将 SoundTransform 对象应用于单个 SoundChannel 对象、全局 SoundMixer 对象或 Microphone 对象等。
flash.media.ID3Info	ID3Info 对象包含一些属性，它们表示通常存储在 mp3 声音文件中的 ID3 元数据信息。
flash.media.Microphone	Microphone 类表示连接到用户计算机上的麦克风或其它声音输入设备。可以将来自麦克风的音频输入传送到本地扬声器或发送到远程服务器。Microphone 对象控制其自己的声音流的增益、采样率以及其它特性。

加载和播放的每种声音需要其自己的 Sound 类和 SoundChannel 类的实例。然后，全局 SoundMixer 类在回放期间将来自多个 SoundChannel 实例的输出混合在一起。

Sound、 SoundChannel 和 SoundMixer 类不能用于从麦克风或流媒体服务器（如 Flash Media Server）中获取的声音数据。

加载外部声音文件

Sound 类的每个实例可加载并触发特定声音资源的回放。应用程序无法重复使用 Sound 对象来加载多种声音。如果它要加载新的声音资源，则应创建一个新的 Sound 对象。

如果要加载较小的声音文件（如要附加到按钮上的单击声音），应用程序可以创建一个新的 Sound，并让其自动加载该声音文件，如下所示：

```
var req:URLRequest = new URLRequest("click.mp3");
var s:Sound = new Sound(req);
```

Sound() 构造函数接受一个 URLRequest 对象作为其第一个参数。当提供 URLRequest 参数的值后，新的 Sound 对象将自动开始加载指定的声音资源。

除了最简单的情况外，应用程序都应关注声音的加载进度，并监视在加载期间出现的错误。例如，如果单击声音非常大，在用户单击触发该声音的按钮时，该声音可能没有完全加载。尝试播放未加载的声音可能会导致运行时错误。较为稳妥的做法是等待声音完全加载后，再让用户执行可能启动声音播放的动作。

Sound 对象将在声音加载过程中调度多种不同的事件。应用程序可以侦听这些事件以跟踪加载进度，并确保在播放之前完全加载声音。下表列出了可以由 Sound 对象调度的事件。

事件	描述
open (Event.OPEN)	就在声音加载操作开始之前进行调度。
progress (ProgressEvent.PROGRESS)	从文件或流接收数据时，在声音加载过程中定期进行调度。
id3 (Event.ID3)	当存在可用于 mp3 声音的 ID3 数据时进行调度。
complete (Event.COMPLETE)	在加载了所有声音资源的数据后进行调度。
ioError (IOErrorEvent.IO_ERROR)	在以下情况下进行调度：找不到声音文件，或者在收到所有声音数据之前加载过程中断。

以下代码说明了如何在完成加载后播放声音：

```
import flash.events.Event;
import flash.media.Sound;
import flash.net.URLRequest;

var s:Sound = new Sound();
s.addEventListener(Event.COMPLETE, onSoundLoaded);
var req:URLRequest = new URLRequest("bigSound.mp3");
s.load(req);

function onSoundLoaded(event:Event):void
{
    var localSound:Sound = event.target as Sound;
    localSound.play();
}
```

首先，该代码范例创建一个新的 Sound 对象，但没有为其指定 URLRequest 参数的初始值。然后，它通过 Sound 对象侦听 Event.COMPLETE 事件，该对象导致在加载完所有声音数据后执行 onSoundLoaded() 方法。接下来，它使用新的 URLRequest 值为声音文件调用 Sound.load() 方法。

在加载完声音后，将执行 onSoundLoaded() 方法。Event 对象的 target 属性是对 Sound 对象的引用。如果调用 Sound 对象的 play() 方法，则会启动声音播放。

监视声音加载过程

声音文件可能很大，而需要花很长时间进行加载。尽管 Flash Player 和 AIR 允许应用程序甚至可以在完全加载声音之前播放声音，但您可能希望向用户指示已加载了多少声音数据以及已播放了多少声音。

Sound 类调度以下两个事件，它们可使声音加载进度显示变得相对比较简单：ProgressEvent.PROGRESS 和 Event.COMPLETE。以下示例说明了如何使用这些事件来显示有关所加载的声音的进度信息：

```
import flash.events.Event;
import flash.events.ProgressEvent;
import flash.media.Sound;
import flash.net.URLRequest;

var s:Sound = new Sound();
s.addEventListener(ProgressEvent.PROGRESS, onLoadProgress);
s.addEventListener(Event.COMPLETE, onLoadComplete);
s.addEventListener(IOErrorEvent.IO_ERROR, onIOError);

var req:URLRequest = new URLRequest("bigSound.mp3");
s.load(req);

function onLoadProgress(event:ProgressEvent):void
{
    var loadedPct:uint = Math.round(100 * (event.bytesLoaded / event.bytesTotal));
    trace("The sound is " + loadedPct + "% loaded.");
}

function onLoadComplete(event:Event):void
{
    var localSound:Sound = event.target as Sound;
    localSound.play();
}
function onIOError(event:IOErrorEvent)
{
    trace("The sound could not be loaded: " + event.text);
}
```

此代码先创建一个 **Sound** 对象，然后向该对象添加侦听器以侦听 **ProgressEvent.PROGRESS** 和 **Event.COMPLETE** 事件。在调用 **Sound.load()** 方法并从声音文件接收第一批数据后，将会发生 **ProgressEvent.PROGRESS** 事件并触发 **onLoadProgress()** 方法。

已加载的声音数据百分比等于 **ProgressEvent** 对象的 **bytesLoaded** 属性值除以 **bytesTotal** 属性值。**Sound** 对象上也提供了相同的 **bytesLoaded** 和 **bytesTotal** 属性。以上示例只显示了有关声音加载进度的消息，但您可以方便地使用 **bytesLoaded** 和 **bytesTotal** 值来更新进度栏组件，例如随 **Adobe Flex 3** 框架或 **Flash** 创作工具提供的组件。

此示例还说明了应用程序在加载声音文件时如何识别并响应出现的错误。例如，如果找不到具有给定文件名的声音文件，**Sound** 对象将调度一个 **Event.IO_ERROR** 事件。在上面的代码中，当发生错误时，将执行 **onIOError()** 方法并显示一条简短的错误消息。

处理嵌入的声音

对于在应用程序用户界面中用作指示器的较小声音（如在单击按钮时播放的声音），最适合使用嵌入的声音（而不是从外部文件加载声音）。

在应用程序中嵌入声音文件时，生成的 **SWF** 文件大小比原来增加了声音文件的大小。也就是说，如果在应用程序中嵌入较大的声音文件，可能会使 **SWF** 文件增大到难以接受的大小。

将声音文件嵌入到应用程序的 **SWF** 文件中的具体方法因开发环境而异。

在 Flash 中使用嵌入的声音文件

Flash 创作工具可导入多种声音格式的声音并将其作为元件存储在库中。然后，您可以将其分配给时间轴上的帧或按钮状态的帧，通过行为来使用声音，或直接在 **ActionScript** 代码中使用它们。本节说明如何在 **ActionScript** 代码中通过 **Flash** 创作工具来使用嵌入的声音。有关如何以其它方法在 **Flash** 中使用嵌入的声音的信息，请参阅《使用 Flash》中的“导入声音”。

使用 **Flash** 创作工具嵌入声音文件：

- 1 选择“文件”>“导入”>“导入到库”，然后选择一个声音文件并导入它。
- 2 在“库”面板中，右键单击导入的文件的名称，然后选择“属性”。单击“为 ActionScript 导出”复选框。
- 3 在“类”字段中，输入一个名称，以便在 ActionScript 中引用此嵌入的声音时使用。默认情况下，它将使用此字段中声音文件的名称。如果文件名包含句点（如名称“DrumSound.mp3”），则必须将其更改为类似于“DrumSound”这样的名称；ActionScript 不允许在类名称中出现句点字符。“基类”字段应仍显示 flash.media.Sound。
- 4 单击“确定”。可能出现一个对话框，指出无法在类路径中找到该类的定义。单击“确定”以继续。如果输入的类名称与应用程序的类路径中任何类的名称都不匹配，则会自动生成从 flash.media.Sound 类继承的新类。
- 5 要使用嵌入的声音，请在 ActionScript 中引用该声音的类名称。例如，通过创建自动生成的 DrumSound 类的一个新实例来启动以下代码：

```
var drum:DrumSound = new DrumSound();
var channel:SoundChannel = drum.play();
```

DrumSound 是 flash.media.Sound 类的子类，所以它继承了 Sound 类的方法和属性，包括上面显示的 play() 方法。

处理声音流文件

如果在仍加载声音文件或视频文件的数据的同时播放该文件，则认为是流式传输。通常，将对从远程服务器加载的外部声音文件进行流式传输，以使用户不必等待加载完所有声音数据再收听声音。

SoundMixer.bufferTime 属性表示 Flash Player 或 AIR 在允许播放声音之前应收集多长时间的声音数据（以毫秒为单位）。也就是说，如果将 bufferTime 属性设置为 5000，则在开始播放声音之前，Flash Player 或 AIR 会从声音文件中加载至少相当于 5000 毫秒的数据。SoundMixer.bufferTime 默认值为 1000。

通过在加载声音时显式地指定新的 bufferTime 值，应用程序可以覆盖单个声音的全局 SoundMixer.bufferTime 值。要覆盖默认缓冲时间，请先创建一个新的 SoundLoaderContext 类实例，设置其 bufferTime 属性，然后将其作为参数传递给 Sound.load() 方法，如下所示：

```
import flash.media.Sound;
import flash.media.SoundLoaderContext;
import flash.net.URLRequest;

var s:Sound = new Sound();
var req:URLRequest = new URLRequest("bigSound.mp3");
var context:SoundLoaderContext = new SoundLoaderContext(8000, true);
s.load(req, context);
s.play();
```

当播放继续进行时，Flash Player 和 AIR 尝试将声音缓冲保持在相同大小或更大。如果声音数据的加载速度比回放快，回放将继续进行而不会中断。但是，如果数据加载速率由于网络限制而减慢，播放头可能会到达声音缓冲区的结尾。如果发生这种情况，将暂停回放，但会在加载更多声音数据后自动恢复回放。

若要查明暂停播放是否是由于 Flash Player 或 AIR 正在等待加载数据造成的，请使用 Sound.isBuffering 属性。

处理动态生成的音频

注：从 Flash Player 10 和 Adobe AIR 1.5 开始可以动态生成音频。

可以动态生成音频数据，而不是加载或流式传输现有声音。在为 Sound 对象的 sampleData 事件分配事件侦听器时，可以生成音频数据。（sampleData 事件在 flash.events 包的 SampleDataEvent 类中定义。）在这种情况下，Sound 对象不从文件中加载声音数据。相反，该对象将用作声音数据的套接字，声音数据通过使用您分配给此事件的函数流入该对象。

向 Sound 对象添加 sampleData 事件侦听器时，该对象会定期请求要添加到声音缓冲区的数据。此缓冲区包含要播放的 Sound 对象的数据。调用 Sound 对象的 play() 方法时，该方法将在请求新声音数据时调度 sampleData 事件。（仅当 Sound 对象尚未从文件中加载 mp3 数据时，这一点才适用。）

SampleDataEvent 对象包含 data 属性。在事件侦听器中，将 ByteArray 对象写入此 data 对象。写入此对象的字节数组将添加到 Sound 对象播放的缓冲声音数据。该缓冲区中的字节数组是一系列介于 -1 和 1 之间的浮点值。每个浮点值均表示声音样本的一个声道（左声道或右声道）的幅值。声音的采样率为每秒 44,100 个样本。每个样本均包含一个左声道和一个右声道，它们交替作为字节数组中的浮点数据。

在处理函数中，使用 ByteArray.writeFloat() 方法写入 sampleData 事件的 data 属性。例如，下面的代码会生成一个正弦波：

```
var mySound:Sound = new Sound();
mySound.addEventListener(SampleDataEvent.SAMPLE_DATA, sineWaveGenerator);
mySound.play();
function sineWaveGenerator(event:SampleDataEvent):void
{
    for (var i:int = 0; i < 8192; i++)
    {
        var n:Number = Math.sin((i + event.position) / Math.PI / 4);
        event.data.writeFloat(n);
        event.data.writeFloat(n);
    }
}
```

调用 Sound.play() 时，应用程序开始调用事件处理函数，请求声音样本数据。播放声音时，应用程序继续发送事件，直至停止提供数据或调用 SoundChannel.stop() 为止。

事件的滞后时间因平台而异，在以后的 Flash Player 和 AIR 版本中也可能改变。请不要依赖某个特定的滞后时间；而应计算出相应的滞后时间。若要计算滞后时间，请使用以下公式：

```
(SampleDataEvent.position / 44.1) - SoundChannelObject.position
```

为 SampleDataEvent 对象的 data 属性提供 2048 到 8192 个样本（在每次调用事件侦听器时）。为了获得最佳性能，请尽可能多地提供样本（最多为 8192 个）。提供的样本越少，在播放过程中就越有可能出现单击和弹出事件。此行为因平台而异，会在各种情况下发生。例如，当调整浏览器的大小时。如果只提供 2048 个样本，则适用于一个平台的代码可能无法适用于其它平台。若要尽可能缩短滞后时间，请考虑允许用户选择数据量。

如果提供的样本数少于 2048（在每次调用 sampleData 事件侦听器时），应用程序会在播放其余样本后停止。然后，它会调度 SoundComplete 事件。

修改来自 mp3 数据的声音

使用 Sound.extract() 方法可从 Sound 对象中提取数据。可以使用（和修改）这些数据，以将其写入要播放的另一个 Sound 对象的动态流。例如，下面的代码使用已加载的 MP3 文件的字节，通过过滤函数 upOctave() 传递这些字节：

```
var mySound:Sound = new Sound();
var sourceSnd:Sound = new Sound();
var urlReq:URLRequest = new URLRequest("test.mp3");
sourceSnd.load(urlReq);
sourceSnd.addEventListener(Event.COMPLETE, loaded);
function loaded(event:Event):void
{
    mySound.addEventListener(SampleDataEvent.SAMPLE_DATA, processSound);
    mySound.play();
}
function processSound(event:SampleDataEvent):void
{
    var bytes:ByteArray = new ByteArray();
    sourceSnd.extract(bytes, 8192);
    event.data.writeBytes(upOctave(bytes));
}
function upOctave(bytes:ByteArray):ByteArray
{
    var returnBytes:ByteArray = new ByteArray();
    bytes.position = 0;
    while(bytes.bytesAvailable > 0)
    {
        returnBytes.writeFloat(bytes.readFloat());
        returnBytes.writeFloat(bytes.readFloat());
        if (bytes.bytesAvailable > 0)
        {
            bytes.position += 8;
        }
    }
    return returnBytes;
}
```

对生成的声音的限制

将 sampleData 事件侦听器与 Sound 对象一起使用时，启用的其它 Sound 方法仅包括 Sound.extract() 和 Sound.play()。调用任何其它方法或属性都将导致异常。仍启用 SoundChannel 对象的所有方法和属性。

播放声音

播放加载的声音非常简便，您只需为 Sound 对象调用 Sound.play() 方法，如下所示：

```
var snd:Sound = new Sound(new URLRequest("smallSound.mp3"));
snd.play();
```

使用 ActionScript 3.0 回放声音时，您可以执行以下操作：

- 从特定起始位置播放声音
- 暂停声音并稍后从相同位置恢复回放
- 准确了解何时播放完声音
- 跟踪声音的回放进度
- 在播放声音的同时更改音量或声相

要在回放期间执行这些操作，请使用 SoundChannel、 SoundMixer 和 SoundTransform 类。

SoundChannel 类控制一种声音的回放。可以将 SoundChannel.position 属性视为播放头，以指示所播放的声音数据中的当前位置。

当应用程序调用 Sound.play() 方法时，将创建一个新的 SoundChannel 类实例来控制播放。

通过将特定起始位置（以毫秒为单位）作为 Sound.play() 方法的 startTime 参数进行传递，应用程序可以从该位置播放声音。它也可以通过在 Sound.play() 方法的 loops 参数中传递一个数值，指定快速且连续地将声音重复播放固定的次数。

使用 startTime 参数和 loops 参数调用 Sound.play() 方法时，每次将从相同的起始点重复播放声音，如以下代码中所示：

```
var snd:Sound = new Sound(new URLRequest("repeatingSound.mp3"));
snd.play(1000, 3);
```

在此示例中，从声音开始后的 1 秒起连续播放声音三次。

暂停和恢复播放声音

如果应用程序播放很长的声音（如歌曲或播客），您可能需要让用户暂停和恢复回放这些声音。实际上，无法在 ActionScript 中的回放期间暂停声音；而只能将其停止。但是，可以从任何位置开始播放声音。您可以记录声音停止时的位置，并随后从该位置开始重放声音。

例如，假定代码加载并播放一个声音文件，如下所示：

```
var snd:Sound = new Sound(new URLRequest("bigSound.mp3"));
var channel:SoundChannel = snd.play();
```

在播放声音的同时，SoundChannel.position 属性指示当前播放到的声音文件位置。应用程序可以在停止播放声音之前存储位置值，如下所示：

```
var pausePosition:int = channel.position;
channel.stop();
```

要恢复播放声音，请传递以前存储的位置值，以便从声音以前停止的相同位置重新启动声音。

```
channel = snd.play(pausePosition);
```

监视回放

应用程序可能需要了解何时停止播放某种声音，以便开始播放另一种声音，或者清除在以前回放期间使用的某些资源。

SoundChannel 类在其声音完成播放时将调度 Event.SOUND_COMPLETE 事件。应用程序可以侦听此事件并执行相应的动作，如下所示：

```
import flash.events.Event;
import flash.media.Sound;
import flash.net.URLRequest;

var snd:Sound = new Sound();
var req:URLRequest = new URLRequest("smallSound.mp3");
snd.load(req);

var channel:SoundChannel = snd.play();
channel.addEventListener(Event.SOUND_COMPLETE, onPlaybackComplete);

public function onPlaybackComplete(event:Event)
{
    trace("The sound has finished playing.");
}
```

SoundChannel 类在回放期间不调度进度事件。要报告回放进度，应用程序可以设置其自己的计时机制并跟踪声音播放头的位置。

若要计算已播放的声音百分比，您可以将 SoundChannel.position 属性值除以所播放的声音数据长度：

```
var playbackPercent:uint = 100 * (channel.position / snd.length);
```

但是，仅当在开始回放之前完全加载了声音数据，此代码才会报告精确的回放百分比。Sound.length 属性显示当前加载的声音数据的大小，而不是整个声音文件的最终大小。要跟踪仍在加载的声音流的回放进度，应用程序应估计完整声音文件的最终大小，并在其计算中使用该值。您可以使用 Sound 对象的 bytesLoaded 和 bytesTotal 属性来估计声音数据的最终长度，如下所示：

```
var estimatedLength:int =
    Math.ceil(snd.length / (snd.bytesLoaded / snd.bytesTotal));
var playbackPercent:uint = 100 * (channel.position / estimatedLength);
```

下面的代码加载一个较大的声音文件，并使用 Event.ENTER_FRAME 事件作为其计时机制来显示播放进度。它定期报告回放百分比，这是作为当前位置值除以声音数据的总长度来计算的：

```
import flash.events.Event;
import flash.media.Sound;
import flash.net.URLRequest;

var snd:Sound = new Sound();
var req:URLRequest = new URLRequest("http://av.adobe.com/podcast/csbu_dev_podcast_epi_2.mp3");
snd.load(req);

var channel:SoundChannel;
channel = snd.play();
addEventListerner(Event.ENTER_FRAME, onEnterFrame);
channel.addEventListerner(Event.SOUND_COMPLETE, onPlaybackComplete);

function onEnterFrame(event:Event):void
{
    var estimatedLength:int =
        Math.ceil(snd.length / (snd.bytesLoaded / snd.bytesTotal));
    var playbackPercent:uint =
        Math.round(100 * (channel.position / estimatedLength));
    trace("Sound playback is " + playbackPercent + "% complete.");
}

function onPlaybackComplete(event:Event)
{
    trace("The sound has finished playing.");
    removeEventListerner(Event.ENTER_FRAME, onEnterFrame);
}
```

在开始加载声音数据后，此代码会调用 snd.play() 方法，并将生成的 SoundChannel 对象存储在 channel 变量中。随后，此代码在主应用程序中添加 Event.ENTER_FRAME 事件的事件侦听器，并在 SoundChannel 对象中添加另一个事件侦听器，用于侦听在播放完成时发生的 Event.SOUND_COMPLETE 事件。

每次应用程序到达其动画中的新帧时，将调用 onEnterFrame() 方法。onEnterFrame() 方法基于已加载的数据量来估计声音文件的总长度，然后计算并显示当前播放百分比。

当播放完整个声音后，将执行 onPlaybackComplete() 方法来移除 Event.ENTER_FRAME 事件的事件侦听器，以使其在完成播放后不会尝试显示进度更新。

可以每秒多次调度 Event.ENTER_FRAME 事件。在某些情况下，您不需要频繁显示播放进度。在这些情况下，应用程序可以使用 flash.util.Timer 类来设置其自己的计时机制；请参阅第 108 页的“[处理日期和时间](#)”。

停止声音流

在进行流式传输的声音（即，在播放的同时仍在加载声音）的播放过程中，有一个奇怪的现象。当应用程序对播放声音流的 SoundChannel 实例调用 SoundChannel.stop() 方法时，声音播放在一个帧处停止，随后在下一帧处从声音开头重新播放。发生这种情况是因为，声音加载过程仍在进行当中。若要同时停止声音流加载和播放，请调用 Sound.close() 方法。

加载和播放声音时的安全注意事项

可以根据 Flash Player 或 AIR 安全模型来限制应用程序访问声音数据的能力。每种声音受两种不同的安全沙箱的限制：内容本身的沙箱（“内容沙箱”）以及加载和播放声音的应用程序或对象的沙箱（“所有者沙箱”）。对于应用程序安全沙箱中的 AIR 应用程序内容，应用程序安全沙箱中的内容可以访问所有声音（包括从其它域加载的声音）。但是，其它安全沙箱中的内容与在 Flash Player 中运行的内容遵守相同的规则。有关 Flash Player 安全模型概述的详细信息以及沙箱定义，请参阅第 587 页的“[Flash Player 安全性](#)”。

内容沙箱控制使用 id3 属性还是 SoundMixer.computeSpectrum() 方法从声音提取详细声音数据。它不会限制声音文件本身的加载或播放。

声音文件的原始域定义了内容沙箱的安全限制。一般来说，如果某个声音文件与加载该文件的应用程序或对象的 SWF 文件位于相同的域或文件夹中，则应用程序或对象具有该声音文件的完全访问权限。如果声音来自不同于应用程序所在域的域，仍可以使用策略文件将其加载到内容沙箱中。

应用程序可以将带有 checkPolicyFile 属性的 SoundLoaderContext 对象作为参数传递给 Sound.load() 方法。如果将 checkPolicyFile 属性设置为 true，则会通知 Flash Player 或 AIR 在从中加载声音的服务器上查找策略文件。如果存在策略文件，并且它为执行加载的 SWF 文件所在的域授予了访问权限，则该 SWF 文件可以加载声音文件，访问 Sound 对象的 id3 属性以及为加载的声音调用 SoundMixer.computeSpectrum() 方法。

所有者沙箱控制声音的本地回放。所有者沙箱是由开始播放声音的应用程序或对象定义的。

只要当前播放的所有 SoundChannel 对象中的声音符合以下条件， SoundMixer.stopAll() 方法就会将它们静音：

- 声音是由相同所有者沙箱中的对象启动的。
- 声音来自具有某一策略文件的源，该策略文件为调用 SoundMixer.stopAll() 方法的应用程序或对象所在的域授予访问权限。

但在 AIR 应用程序中，应用程序安全沙箱中的内容（随 AIR 应用程序安装的内容）不受这些安全限制的约束。

要查明 SoundMixer.stopAll() 方法是否确实停止了所有播放的声音，应用程序可以调用 SoundMixer.areSoundsInaccessible() 方法。如果该方法返回值 true，则当前所有者沙箱无法控制播放的某些声音， SoundMixer.stopAll() 方法不会将其停止。

SoundMixer.stopAll() 方法还会阻止播放头继续播放从外部文件加载的所有声音。但是，如果动画移动到一个新帧， FLA 文件中嵌入的声音以及使用 Flash 创作工具附加到时间轴中的帧上的声音可能会重新开始播放。

控制音量和声相

单个 SoundChannel 对象控制声音的左和右立体声声道。如果 mp3 声音是单声道声音， SoundChannel 对象的左和右立体声声道将包含完全相同的波形。

可通过使用 SoundChannel 对象的 leftPeak 和 rightPeak 属性来查明所播放的声音的每个立体声声道的波幅。这些属性显示声音波形本身的峰值波幅。它们并不表示实际回放音量。实际回放音量是声音波形的波幅以及 SoundChannel 对象和 SoundMixer 类中设置的音量值的函数。

在回放期间，可以使用 SoundChannel 对象的 pan 属性为左和右声道分别指定不同的音量级别。pan 属性可以具有范围从 -1 到 1 的值，其中， -1 表示左声道以最大音量播放，而右声道处于静音状态； 1 表示右声道以最大音量播放，而左声道处于静音状态。介于 -1 和 1 之间的数值为左和右声道值设置一定比例的值，值 0 表示两个声道以均衡的中音量级别播放。

以下代码示例使用 volume 值 0.6 和 pan 值 -1 创建一个 SoundTransform 对象（左声道为最高音量，右声道没有音量）。此代码将 SoundTransform 对象作为参数传递给 play() 方法，此方法将该 SoundTransform 对象应用于为控制播放而创建的新 SoundChannel 对象。

```
var snd:Sound = new Sound(new URLRequest("bigSound.mp3"));
var trans:SoundTransform = new SoundTransform(0.6, -1);
var channel:SoundChannel = snd.play(0, 1, trans);
```

可以在播放声音的同时更改音量和声相控制，方法是设置 SoundTransform 对象的 pan 或 volume 属性，然后将该对象作为 SoundChannel 对象的 soundTransform 属性进行应用。

也可以通过使用 SoundMixer 类的 soundTransform 属性，同时为所有声音设置全局音量和声相值，如以下示例所示：

```
SoundMixer.soundTransform = new SoundTransform(1, -1);
```

也可以使用 SoundTransform 对象为 Microphone 对象设置 volume 和 pan 值（请参阅第 488 页的“[捕获声音输入](#)”），并可以为 Sprite 对象和 SimpleButton 对象设置这些值。

以下示例在播放声音的同时将声音从左声道移到右声道，然后再移回来，并交替进行这一过程。

```
import flash.events.Event;
import flash.media.Sound;
import flash.media.SoundChannel;
import flash.media.SoundMixer;
import flash.net.URLRequest;

var snd:Sound = new Sound();
var req:URLRequest = new URLRequest("bigSound.mp3");
snd.load(req);

var panCounter:Number = 0;

var trans:SoundTransform;
trans = new SoundTransform(1, 0);
var channel:SoundChannel = snd.play(0, 1, trans);
channel.addEventListener(Event.SOUND_COMPLETE, onPlaybackComplete);

addEventListener(Event.ENTER_FRAME, onEnterFrame);

function onEnterFrame(event:Event):void
{
    trans.pan = Math.sin(panCounter);
    channel.soundTransform = trans; // or SoundMixer.soundTransform = trans;
    panCounter += 0.05;
}

function onPlaybackComplete(event:Event):void
{
    removeEventListener(Event.ENTER_FRAME, onEnterFrame);
}
```

此代码先加载一个声音文件，然后将 volume 设置为 1（最大音量）并将 pan 设置为 0（声音在左和右声道之间均衡地平均分布）以创建一个新的 SoundTransform 对象。接下来，此代码调用 snd.play() 方法，以将 SoundTransform 对象作为参数进行传递。

在播放声音时，将反复执行 onEnterFrame() 方法。onEnterFrame() 方法使用 Math.sin() 函数来生成介于 -1 和 1 之间的值，此范围对应于可接受的 SoundTransform.pan 属性值。此代码将 SoundTransform 对象的 pan 属性设置为新值，然后设置声道的 soundTransform 属性以使用更改后的 SoundTransform 对象。

要运行此示例，请用本地 mp3 文件的名称替换文件名 bigSound.mp3。然后，运行该示例。当右声道音量变小时，您应会听到左声道音量变大，反之亦然。

在此示例中，可通过设置 SoundMixer 类的 soundTransform 属性来获得同样的效果。但是，这会影响当前播放的所有声音的声相，而不是只影响此 SoundChannel 对象播放的一种声音。

处理声音元数据

使用 mp3 格式的声音文件可以采用 ID3 标签格式来包含有关声音的其它数据。

并非每个 mp3 文件都包含 ID3 元数据。当 Sound 对象加载 mp3 声音文件时，如果该声音文件包含 ID3 元数据，它将调度 Event.ID3 事件。若要防止出现运行时错误，应用程序应等待接收 Event.ID3 事件后，再访问加载的声音的 Sound.id3 属性。

以下代码说明了如何识别何时加载了声音文件的 ID3 元数据：

```
import flash.events.Event;
import flash.media.ID3Info;
import flash.media.Sound;

var s:Sound = new Sound();
s.addEventListener(Event.ID3, onID3InfoReceived);
s.load("mySound.mp3");

function onID3InfoReceived(event:Event)
{
    var id3:ID3Info = event.target.id3;

    trace("Received ID3 Info:");
    for (var propName:String in id3)
    {
        trace(propName + " = " + id3[propName]);
    }
}
```


此代码先创建一个 Sound 对象并通知该对象侦听 Event.ID3 事件。加载了声音文件的 ID3 元数据后，将调用 onID3InfoReceived() 方法。传递给 onID3InfoReceived() 方法的 Event 对象的目标是原始 Sound 对象，因此，该方法随后获取 Sound 对象的 id3 属性，然后循环访问其所有命名属性以跟踪这些属性的值。

访问原始声音数据

通过使用 SoundMixer.computeSpectrum() 方法，应用程序可以读取当前所播放的波形的原始声音数据。如果当前播放多个 SoundChannel 对象， SoundMixer.computeSpectrum() 方法将显示混合在一起的每个 SoundChannel 对象的组合声音数据。声音数据是作为 `ByteArray` 对象（包含 512 个字节的数据）返回的，其中的每个字节包含一个介于 -1 和 1 之间的浮点值。这些值表示所播放的声音波形中的点的波幅。这些值是分为两个组（每组包含 256 个值）提供的，第一个组用于左立体声声道，第二个组用于右立体声声道。

如果将 `FFTMode` 参数设置为 `true`， `SoundMixer.computeSpectrum()` 方法将返回频谱数据，而非波形数据。频谱显示按声音频率（从最低频率到最高频率）排列的波幅。可以使用快速傅立叶变换（FFT）将波形数据转换为频谱数据。生成的频谱值范围介于 0 和约 1.414（2 的平方根）之间。

下图比较了将 `FFTMode` 参数设置为 `true` 和 `false` 时从 `computeSpectrum()` 方法返回的数据。此图所用数据的声音在左声道中包含很大的低音；而在右声道中包含击鼓声。



SoundMixer.computeSpectrum() 方法返回的值
A. fftMode=true B. fftMode=false

`computeSpectrum()` 方法也可以返回已在较低比特率重新采样的数据。通常，这会产生更平滑的波形数据或频率数据，但会以牺牲细节为代价。`stretchFactor` 参数控制 `computeSpectrum()` 方法数据的采样率。如果将 `stretchFactor` 参数设置为 0（默认值），则以采样率 44.1 kHz 采集声音数据样本。`stretchFactor` 参数值每连续增加 1，采样率就减小一半，因此，值 1 指定采样率 22.05 kHz，值 2 指定采样率 11.025 kHz，依此类推。当使用较高的 `stretchFactor` 值时，`computeSpectrum()` 方法仍会为每个立体声声道返回 256 个字节。

`SoundMixer.computeSpectrum()` 方法具有一些限制：

- 由于来自麦克风或 RTMP 流的声音数据不是通过全局 `SoundMixer` 对象传递的，因此，`SoundMixer.computeSpectrum()` 方法不会从这些源返回数据。
- 如果播放的一种或多种声音来自当前内容沙箱以外的源，安全限制将导致 `SoundMixer.computeSpectrum()` 方法引发错误。有关 `SoundMixer.computeSpectrum()` 方法的安全限制的更多详细信息，请参阅第 483 页的“[加载和播放声音时的安全注意事项](#)”和第 602 页的“[作为数据访问加载的媒体](#)”。

但在 AIR 应用程序中，应用程序安全沙箱中的内容（随 AIR 应用程序安装的内容）不受这些安全限制的约束。

构建简单的声音可视化程序

以下示例使用 `SoundMixer.computeSpectrum()` 方法来显示声音波形图（它对每个帧进行动画处理）：

```
import flash.display.Graphics;
import flash.events.Event;
import flash.media.Sound;
import flash.media.SoundChannel;
import flash.media.SoundMixer;
import flash.net.URLRequest;

const PLOT_HEIGHT:int = 200;
const CHANNEL_LENGTH:int = 256;

var snd:Sound = new Sound();
var req:URLRequest = new URLRequest("bigSound.mp3");
snd.load(req);

var channel:SoundChannel;
channel = snd.play();
addEventListener(Event.ENTER_FRAME, onEnterFrame);
snd.addEventListener(Event.SOUND_COMPLETE, onPlaybackComplete);

var bytes:ByteArray = new ByteArray();

function onEnterFrame(event:Event):void
{
    SoundMixer.computeSpectrum(bytes, false, 0);

    var g:Graphics = this.graphics;

    g.clear();
    g.lineStyle(0, 0x6600CC);
    g.beginFill(0x6600CC);
    g.moveTo(0, PLOT_HEIGHT);

    var n:Number = 0;

    // left channel
    for (var i:int = 0; i < CHANNEL_LENGTH; i++)
    {
        n = (bytes.readFloat() * PLOT_HEIGHT);
        g.lineTo(i * 2, PLOT_HEIGHT - n);
    }
    g.lineTo(CHANNEL_LENGTH * 2, PLOT_HEIGHT);
    g.endFill();


    // right channel
    g.lineStyle(0, 0xCC0066);
    g.beginFill(0xCC0066, 0.5);
    g.moveTo(CHANNEL_LENGTH * 2, PLOT_HEIGHT);

    for (i = CHANNEL_LENGTH; i > 0; i--)
    {
        n = (bytes.readFloat() * PLOT_HEIGHT);
        g.lineTo(i * 2, PLOT_HEIGHT - n);
    }
    g.lineTo(0, PLOT_HEIGHT);
    g.endFill();
}

function onPlaybackComplete(event:Event)
{
    removeEventListener(Event.ENTER_FRAME, onEnterFrame);
}
```

此示例先加载并播放一个声音文件，然后在播放声音的同时侦听将触发 `onEnterFrame()` 方法的 `Event.ENTER_FRAME` 事件。`onEnterFrame()` 方法先调用 `SoundMixer.computeSpectrum()` 方法，后者将声音波形数据存储在 `bytes ByteArray` 对象中。

声音波形是使用矢量绘图 API 绘制的。`for` 循环将循环访问第一批 256 个数据值（表示左立体声声道），然后使用 `Graphics.lineTo()` 方法绘制一条从每个点到下一个点的直线。第二个 `for` 循环将循环访问下一批 256 个值，此时按相反的顺序（从右到左）对它们进行绘制。生成的波形图可能会产生非常有趣的镜像图像效果，如以下图像中所示。



捕获声音输入

应用程序可通过 `Microphone` 类连接到用户系统上的麦克风或其它声音输入设备，并将输入音频广播到该系统的扬声器，或者将音频数据发送到远程服务器（如 Flash Media Server）。不能从麦克风访问原始音频数据；只能将音频发送到系统的扬声器或将压缩的音频数据发送到远程服务器。可以使用 Speex 或 Nellymoser 编解码器处理发送到远程服务器的数据。（从 Flash Player 10 和 Adobe AIR 1.5 开始支持 Speex 编解码器。）

访问麦克风

`Microphone` 类没有构造函数方法。相反，应使用静态 `Microphone.getMicrophone()` 方法来获取新的 `Microphone` 实例，如下所示：

```
var mic:Microphone = Microphone.getMicrophone();
```

不使用参数调用 `Microphone.getMicrophone()` 方法时，将返回在用户系统上发现的第一个声音输入设备。

系统可能连接了多个声音输入设备。应用程序可以使用 `Microphone.names` 属性来获取所有可用声音输入设备名称的数组。然后，它可以使用 `index` 参数（与数组中的设备名称的索引值相匹配）来调用 `Microphone.getMicrophone()` 方法。

系统可能没有连接麦克风或其它声音输入设备。可以使用 `Microphone.names` 属性或 `Microphone.getMicrophone()` 方法来检查用户是否安装了声音输入设备。如果用户未安装声音输入设备，则 `names` 数组的长度为零，并且 `getMicrophone()` 方法返回值 `null`。

当应用程序调用 `Microphone.getMicrophone()` 方法时，Flash Player 将显示“Flash Player 设置”对话框，它提示用户允许或拒绝 Flash Player 对系统上的摄像头和麦克风的访问。在用户单击此对话框中的“允许”或“拒绝”按钮后，将调度 `StatusEvent`。该 `StatusEvent` 实例的 `code` 属性指示是允许还是拒绝对麦克风的访问，如下例所示：

```
import flash.media.Microphone;

var mic:Microphone = Microphone.getMicrophone();
mic.addEventListener(StatusEvent.STATUS, this.onMicStatus);

function onMicStatus(event:StatusEvent):void
{
    if (event.code == "Microphone.Unmuted")
    {
        trace("Microphone access was allowed.");
    }
    else if (event.code == "Microphone.Muted")
    {
        trace("Microphone access was denied.");
    }
}
```

如果允许访问，则 StatusEvent.code 属性将包含“Microphone.Unmuted”，如果拒绝访问，则该属性包含“Microphone.Muted”。

注：当用户允许或拒绝对麦克风的访问时，Microphone.muted 属性将被分别设置为 true 或 false。但是，在调度 StatusEvent 之前，不会在 Microphone 实例上设置 muted 属性，因此，应用程序还应等待调度 StatusEvent.STATUS 事件后再检查 Microphone.muted 属性。

将麦克风音频传送到本地扬声器

可以使用参数值 true 调用 Microphone.setLoopback() 方法，以将来自麦克风的音频输入传送到本地系统扬声器。

如果将来自本地麦克风的声音传送到本地扬声器，则会存在创建音频回馈循环的风险，这可能会导致非常大的振鸣声，并且可能会损坏声音硬件。使用参数值 true 调用 Microphone.setUseEchoSuppression() 方法可降低发生音频回馈的风险，但不会完全消除该风险。Adobe 建议您始终在调用 Microphone.setLoopback(true) 之前调用 Microphone.setUseEchoSuppression(true)，除非您确信用户使用耳机来播放声音，或者使用除扬声器以外的某种设备。

以下代码说明了如何将来自本地麦克风的音频传送到本地系统扬声器：

```
var mic:Microphone = Microphone.getMicrophone();
mic.setUseEchoSuppression(true);
mic.setLoopBack(true);
```

更改麦克风音频

应用程序可以使用两种方法更改来自麦克风的音频数据。第一，它可以更改输入声音的增益，这会有效地将输入值乘以指定的数值以创建更大或更小的声音。Microphone.gain 属性接受介于 0 和 100 之间的数值（含 0 和 100）。值 50 相当于乘数 1，它指定正常音量。值 0 相当于乘数 0，它可有效地将输入音频静音。大于 50 的值指定的音量高于正常音量。

应用程序也可以更改输入音频的采样率。较高的采样率可提高声音品质，但它们也会创建更密集的数据流（使用更多的资源进行传输和存储）。Microphone.rate 属性表示以千赫 (kHz) 为单位测量的音频采样率。默认采样率是 8 kHz。如果麦克风支持较高的采样率，您可以将 Microphone.rate 属性设置为高于 8 kHz 的值。例如，如果将 Microphone.rate 属性设置为值 11，则会将采样率设置为 11 kHz；如果将该属性设置为 22，则会将采样率设置为 22 kHz，依此类推。可用的采样率取决于所选择的编解码器。使用 Nellymoser 编解码器时，可以将采样率指定为 5、8、11、16、22 和 44 kHz。使用 Speex 编解码器（Flash Player 10 和 Adobe AIR 1.5 中开始提供）时，只能使用 16 kHz。

检测麦克风活动

为节省带宽和处理资源，Flash Player 将尝试检测何时麦克风不传输声音。当麦克风的活动级别处于静音级别阈值以下一段时间后，Flash Player 将停止传输音频输入，并改为调度一个简单的 ActivityEvent。如果使用 Speex 编解码器（Flash Player 10 或更高版本和 Adobe AIR 1.5 或更高版本中提供），请将静音级别设置为 0，以确保应用程序持续传输音频数据。Speex 声音活动检测会自动降低带宽。

Microphone 类的以下三个属性用于监视和控制活动检测：

- activityLevel 只读属性指示麦克风检测的音量，范围从 0 到 100。
- silenceLevel 属性指定激活麦克风并调度 ActivityEvent.ACTIVITY 事件所需的音量。silenceLevel 属性也使用从 0 到 100 的范围，默认值为 10。
- silenceTimeout 属性描述活动级别处于静音级别以下多长时间（以毫秒为单位）后，才会调度 ActivityEvent.ACTIVITY 事件以指示麦克风现在处于静音状态。silenceTimeout 默认值是 2000。

Microphone.silenceLevel 属性和 Microphone.silenceTimeout 属性都是只读的，但可以使用 Microphone.setSilenceLevel() 方法来更改它们的值。

在某些情况下，在检测到新活动时激活麦克风的过程可能会导致短暂的延迟。通过将麦克风始终保持活动状态，可以消除此类激活延迟。应用程序可以调用 Microphone.setSilenceLevel() 方法并将 silenceLevel 参数设置为零，以通知 Flash Player 将麦克风保持活动状态并持续收集音频数据，即使未检测到任何声音也是如此。反之，如果将 silenceLevel 参数设置为 100，则可以完全禁止激活麦克风。

以下示例显示了有关麦克风的信息，并报告 Microphone 对象调度的活动事件和状态事件：

```
import flash.events.ActivityEvent;
import flash.events.StatusEvent;
import flash.media.Microphone;

var deviceArray:Array = Microphone.names;
trace("Available sound input devices:");
for (var i:int = 0; i < deviceArray.length; i++)
{
    trace(" " + deviceArray[i]);
}

var mic:Microphone = Microphone.getMicrophone();
mic.gain = 60;
mic.rate = 11;
mic.setUseEchoSuppression(true);
mic.setLoopBack(true);
mic.setSilenceLevel(5, 1000);

mic.addEventListener(ActivityEvent.ACTIVITY, this.onMicActivity);
mic.addEventListener(StatusEvent.STATUS, this.onMicStatus);
```

```
var micDetails:String = "Sound input device name: " + mic.name + '\n';
micDetails += "Gain: " + mic.gain + '\n';
micDetails += "Rate: " + mic.rate + " kHz" + '\n';
micDetails += "Muted: " + mic.muted + '\n';
micDetails += "Silence level: " + mic.silenceLevel + '\n';
micDetails += "Silence timeout: " + mic.silenceTimeout + '\n';
micDetails += "Echo suppression: " + mic.useEchoSuppression + '\n';
trace(micDetails);

function onMicActivity(event:ActivityEvent):void
{
    trace("activating=" + event.activating + ", activityLevel=" +
          mic.activityLevel);
}

function onMicStatus(event:StatusEvent):void
{
    trace("status: level=" + event.level + ", code=" + event.code);
}
```

在运行上面的示例时，对着系统麦克风说话或发出噪音，并观察所生成的、显示在控制台或调试窗口中的 `trace` 语句。

向媒体服务器发送音频以及从中接收音频

将 ActionScript 与 Flash Media Server 等流媒体服务器配合使用时，可以使用额外的音频功能。

具体而言，应用程序可以将 `Microphone` 对象附加到 `NetStream` 对象，并将数据直接从用户的麦克风传输到服务器。也可以将音频数据从服务器流式传输到用 Flash 或 Flex 创建的应用程序，并将其作为 `MovieClip` 的一部分或使用 `Video` 对象进行播放。

从 Flash Player 10 和 Adobe AIR 1.5 开始提供 Speex 编解码器。若要设置对于发送到媒体服务器的压缩音频所使用的编解码器，请设置 `Microphone` 对象的 `codec` 属性。此属性可以有两个值，这两个值在 `SoundCodec` 类中枚举。如果将 `codec` 属性设置为 `SoundCodec.SPEEX`，则选择 Speex 编解码器处理压缩音频。如果该属性设置为 `SoundCodec.NELLYMOSER`（默认值），则选择 Nellymoser 编解码器处理压缩音频。

有关详细信息，请参阅 Flash Media Server 在线文档，网址为 <http://livedocs.adobe.com>。

示例：Podcast Player

播客是通过 Internet 以按需方式或订阅方式分发的声音文件。播客通常是作为系列的一部分发布的，此系列也称为播客频道。由于播客节目的持续时间从一分钟到数小时不等，因此，通常在播放的同时对其进行流式传输。播客节目（也称为项目）通常是以 `mp3` 文件格式提供的。视频播客也非常受欢迎，但此范例应用程序仅播放使用 `mp3` 文件的音频播客。

此示例并不是一个功能完备的播客聚合器应用程序。例如，它不能管理对特定播客的订阅，或在下次运行应用程序时记住用户已收听的播客。它可用作功能更完备的播客聚合器的起点。

Podcast Player 示例说明了以下 ActionScript 编程方法：

- 读取外部 RSS 新闻频道并分析其 XML 内容
- 创建 `SoundFacade` 类以简化加载和回放声音文件的过程
- 显示声音回放进度
- 暂停和恢复声音回放

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。Podcast Player 应用程序文件位于文件夹 Samples/PodcastPlayer 中。该应用程序包含以下文件：

文件	描述
PodcastPlayer.mxml 或 PodcastPlayer.fla	适用于 Flex (MXML) 或 Flash (FLA) 的应用程序的用户界面。
comp/example/progra mmingas3/podcastplay er/PodcastPlayer.as	包含 podcast 播放器用户界面逻辑的文档类 (仅限 Flash)。
comp/example/progra mmingas3/podcastplay er/SoundPlayer.as	包含声音播放器用户界面逻辑的 SoundPlayer 影片剪辑元件的类 (仅限 Flash)。
comp/example/progra mmingas3/podcastplay er/PlayButtonRenderer .as	用于在数据网格单元格中显示播放按钮的自定义单元格渲染器 (仅限 Flash)。
com/example/program mingas3/podcastplayer /RSSBase.as	为 RSSChannel 类和 RSSItem 类提供公共属性和方法的基类。
com/example/program mingas3/podcastplayer /RSSChannel.as	保存 RSS 频道的相关数据的 ActionScript 类。
com/example/program mingas3/podcastplayer /RSSItem.as	保存 RSS 项目的相关数据的 ActionScript 类。
com/example/program mingas3/podcastplayer /SoundFacade.as	应用程序的主 ActionScript 类。它封装 Sound 类和 SoundChannel 类的方法和事件，并添加对回放暂停和恢复的支持。
com/example/program mingas3/podcastplayer /URLService.as	从远程 URL 检索数据的 ActionScript 类。
playerconfig.xml	这是一个 XML 文件，其中包含表示播客频道的 RSS 新闻频道列表。
comp/example/progra mmingas3/utils/DateU til.as	用于简化日期格式设置的类 (仅限 Flash)。

读取播客频道的 RSS 数据

Podcast Player 应用程序先读取一些播客频道及其节目的相关信息：

- 首先，应用程序读取包含播客频道列表的 XML 配置文件，并向用户显示该频道列表。
- 当用户选择其中的一个播客频道时，应用程序会读取该频道的 RSS 新闻频道并显示频道节目列表。

此示例使用 URLLoader 实用程序类，从远程位置或本地文件检索基于文本的数据。Podcast Player 先创建一个 URLLoader 对象，以便从 playerconfig.xml 文件中获取采用 XML 格式的 RSS 新闻频道列表。接下来，当用户从列表中选择特定新闻频道时，将创建一个新的 URLLoader 对象以从该新闻频道的 URL 中读取 RSS 数据。

使用 SoundFacade 类简化声音加载和回放

ActionScript 3.0 声音体系结构功能强大，但非常复杂。如果应用程序只需要基本的声音加载和回放功能，可使用通过提供一组更简单的方法调用和事件来隐藏某些复杂性的类。在软件设计模式领域中，这样的类称为“外观”。

SoundFacade 类表示用于执行以下任务的单个接口：

- 使用 **Sound** 对象、**SoundLoaderContext** 对象以及 **SoundMixer** 类来加载声音文件
- 使用 **Sound** 对象和 **SoundChannel** 对象来播放声音文件
- 调度回放进度事件
- 使用 **Sound** 对象和 **SoundChannel** 对象来暂停和恢复声音回放

SoundFacade 类尝试以更简化的方式提供 ActionScript **Sound** 类的大部分功能。

以下代码显示了类声明、类属性以及 **SoundFacade()** 构造函数方法：

```
public class SoundFacade extends EventDispatcher
{
    public var s:Sound;
    public var sc:SoundChannel;
    public var url:String;
    public var bufferTime:int = 1000;

    public var isLoaded:Boolean = false;
    public var isReadyToPlay:Boolean = false;
    public var isPlaying:Boolean = false;
    public var isStreaming:Boolean = true;
    public var autoLoad:Boolean = true;
    public var autoPlay:Boolean = true;

    public var pausePosition:int = 0;

    public static const PLAY_PROGRESS:String = "playProgress";
    public var progressInterval:int = 1000;
    public var playTimer:Timer;

    public function SoundFacade(soundUrl:String, autoLoad:Boolean = true,
                               autoPlay:Boolean = true, streaming:Boolean = true,
                               bufferTime:int = -1):void
    {
        this.url = soundUrl;

        // Sets Boolean values that determine the behavior of this object
        this.autoLoad = autoLoad;
        this.autoPlay = autoPlay;
        this.isStreaming = streaming;

        // Defaults to the global bufferTime value
        if (bufferTime < 0)
        {
            bufferTime = SoundMixer.bufferTime;
        }

        // Keeps buffer time reasonable, between 0 and 30 seconds
        this.bufferTime = Math.min(Math.max(0, bufferTime), 30000);

        if (autoLoad)
        {
            load();
        }
    }
}
```

SoundFacade 类扩展了 **EventDispatcher** 类，以使其能够调度自己的事件。类代码先声明 **Sound** 对象和 **SoundChannel** 对象的属性。该类还会存储声音文件 URL 的值以及对声音进行流式传输时使用的 **bufferTime** 属性。此外，它还接受某些影响加载和回放行为的布尔参数值：

- **autoLoad** 参数通知对象，应在创建此对象后立即启动声音加载。
- **autoPlay** 参数指示在加载了足够多的声音数据后应立即启动声音播放。如果这是声音流，在加载了足够多的数据（由 **bufferTime** 属性指定）后将立即开始播放。
- **streaming** 参数指示可以在加载完成之前开始播放此声音文件。

bufferTime 参数的默认值为 -1。如果构造函数方法在 **bufferTime** 参数中检测到负值，它会将 **bufferTime** 属性设置为 **SoundMixer.bufferTime** 的值。这样，应用程序便可根据需要默认使用全局 **SoundMixer.bufferTime** 值。

如果将 **autoLoad** 参数设置为 **true**，构造函数方法将立即调用以下 **load()** 方法来开始加载声音文件：

```
public function load():void
{
    if (this.isPlaying)
    {
        this.stop();
        this.s.close();
    }
    this.isLoaded = false;

    this.s = new Sound();

    this.s.addEventListener(ProgressEvent.PROGRESS, onLoadProgress);
    this.s.addEventListener(Event.OPEN, onLoadOpen);
    this.s.addEventListener(Event.COMPLETE, onLoadComplete);
    this.s.addEventListener(Event.ID3, onID3);
    this.s.addEventListener(IOErrorEvent.IO_ERROR, onIOError);
    this.s.addEventListener(SecurityErrorEvent.SECURITY_ERROR, onIOError);

    var req:URLRequest = new URLRequest(this.url);

    var context:SoundLoaderContext = new SoundLoaderContext(this.bufferTime, true);
    this.s.load(req, context);
}
```

load() 方法创建一个新的 **Sound** 对象，然后为所有重要的声音事件添加侦听器。接下来，它通知 **Sound** 对象使用 **SoundLoaderContext** 对象传入 **bufferTime** 值以加载声音文件。

由于可以更改 **url** 属性，因此，可以使用 **SoundFacade** 实例来连续播放不同的声音文件：只需更改 **url** 属性并调用 **load()** 方法，即可加载新的声音文件。

以下三个事件侦听器方法说明了 **SoundFacade** 对象如何跟踪加载进度并确定何时开始播放声音：

```

public function onLoadOpen(event:Event):void
{
    if (this.isStreaming)
    {
        this.isReadyToPlay = true;
        if (autoPlay)
        {
            this.play();
        }
    }
    this.dispatchEvent(event.clone());
}

public function onLoadProgress(event:ProgressEvent):void
{
    this.dispatchEvent(event.clone());
}

public function onLoadComplete(event:Event):void
{
    this.isReadyToPlay = true;
    this.isLoaded = true;
    this.dispatchEvent(evt.clone());

    if (autoPlay && !isPlaying)
    {
        play();
    }
}

```

在开始加载声音时，将执行 `onLoadOpen()` 方法。如果可以使用流模式播放声音，`onLoadComplete()` 方法会立即将 `isReadyToPlay` 标志设置为 `true`。`isReadyToPlay` 标志确定应用程序能否开始播放声音，这可能为了响应用户动作，如单击“播放”按钮。`SoundChannel` 类管理声音数据的缓冲，因此在调用 `play()` 方法之前，不需要显式地检查是否加载了足够多的数据。

在加载过程中，将定期执行 `onLoadProgress()` 方法。它仅调度其 `ProgressEvent` 对象的克隆，该对象由使用此 `SoundFacade` 对象的代码使用。

当完全加载了声音数据后，将执行 `onLoadComplete()` 方法，以便为非声音流调用 `play()` 方法（如果需要）。下面显示了 `play()` 方法本身。

```

public function play(pos:int = 0):void
{
    if (!this.isPlaying)
    {
        if (this.isReadyToPlay)
        {
            this.sc = this.s.play(pos);
            this.sc.addEventListener(Event.SOUND_COMPLETE, onPlayComplete);
            this.isPlaying = true;

            this.playTimer = new Timer(this.progressInterval);
            this.playTimer.addEventListener(TimerEvent.TIMER, onPlayTimer);
            this.playTimer.start();
        }
    }
}

```

如果已准备好播放声音，`play()` 方法将调用 `Sound.play()` 方法。生成的 `SoundChannel` 对象存储在 `sc` 属性中。`play()` 方法随后创建一个 `Timer` 对象，该对象用于按固定间隔调度播放进度事件。

显示回放进度

创建 Timer 对象以实现回放监视是一个很复杂的操作，您只需对其编码一次。通过在可重用的类（如 SoundFacade 类）中封装此 Timer 逻辑，应用程序可以在加载和播放声音时侦听相同类型的进度事件。

由 SoundFacade.play() 方法创建的 Timer 对象每秒调度一个 TimerEvent 实例。每当新的 TimerEvent 到达时，就会执行以下 onPlayTimer() 方法：

```
public function onPlayTimer(event:TimerEvent):void
{
    var estimatedLength:int =
        Math.ceil(this.s.length / (this.s.bytesLoaded / this.s.bytesTotal));
    var progEvent:ProgressEvent =
        new ProgressEvent(PLAY_PROGRESS, false, false, this.sc.position, estimatedLength);
    this.dispatchEvent(progEvent);
}
```

onPlayTimer() 方法可实现在第 481 页的“[监视回放](#)”一节中介绍的大小估计技术。然后，它创建一个事件类型为 SoundFacade.PLAY_PROGRESS 的新 ProgressEvent 实例，并将 bytesLoaded 属性设置 SoundChannel 对象的当前位置，而将 bytesTotal 属性设置为估计的声音数据长度。

暂停和恢复回放

以前显示的 SoundFacade.play() 方法接受 pos 参数，该参数与声音数据中的起始位置相对应。如果 pos 值为零，则从开头开始播放声音。

SoundFacade.stop() 方法也接受 pos 参数，如下所示：

```
public function stop(pos:int = 0):void
{
    if (this.isPlaying)
    {
        this.pausePosition = pos;
        this.sc.stop();
        this.playTimer.stop();
        this.isPlaying = false;
    }
}
```

每当调用 SoundFacade.stop() 方法时，它都会设置 pausePosition 属性，以便当用户要恢复播放相同的声音时应用程序知道将播放头放置在什么位置。

下面显示的 SoundFacade.pause() 和 SoundFacade.resume() 方法分别调用 SoundFacade.stop() 和 SoundFacade.play() 方法，以便每次传递 pos 参数值。

```
public function pause():void
{
    stop(this.sc.position);
}

public function resume():void
{
    play(this.pausePosition);
}
```

pause() 方法将当前的 SoundChannel.position 值传递给 play() 方法，后者将该值存储在 pausePosition 属性中。resume() 方法通过使用 pausePosition 值作为起点再次开始播放相同的声音。

扩展 Podcast Player 示例

此示例呈现了一个框架 Podcast Player，用于说明如何使用可重用的 SoundFacade 类。您可以添加其它功能以改进此应用程序的用途，其中包括以下功能：

- 将新闻频道列表和有关每个节目的使用信息存储在 SharedObject 实例中，下次用户运行应用程序时便可以使用它们。
- 允许用户将其自己的 RSS 新闻频道添加到播客频道列表中。
- 当用户停止或离开节目时，记住播放头的位置，以便下次用户运行应用程序时能够从该位置重新开始播放。
- 下载节目的 mp3 文件，以便用户在没有连接到 Internet 时可以脱机收听。
- 添加订阅功能，以便定期检查播客频道中的新节目并自动更新节目列表。
- 使用来自播客托管服务（如 Odeo.com）的 API 添加播客搜索和浏览功能。

第 26 章：捕获用户输入

本章介绍了如何使用 ActionScript 3.0 来创建交互性以响应用户活动。它讨论了键盘和鼠标事件，然后继续介绍更高级的主题，其中包括上下文菜单自定义和焦点管理。本章最后介绍了 WordSearch，这是一个响应鼠标输入的应用程序示例。

请注意，本章假定您已熟悉了 ActionScript 3.0 事件模型。有关详细信息，请参阅第 207 页的“[处理事件](#)”。

用户输入基础知识

捕获用户输入简介

用户交互（无论是通过键盘、鼠标、摄像头还是这些设备的组合）是交互性的基础。在 ActionScript 3.0 中，识别和响应用户交互主要涉及事件侦听。

InteractiveObject 类是 DisplayObject 类的一个子类，它提供了处理用户交互所需的事件和功能的通用结构。您无法直接创建 InteractiveObject 类的实例。而是由显示对象（如 SimpleButton、Sprite、TextField 以及各种 Flash 创作工具和 Flex 组件）从此类中继承其用户交互模型，因而它们共享同一个通用结构。这意味着，您为处理从 InteractiveObject 派生的一个对象中的用户交互而编写的代码以及学会的方法适用于所有其它对象。

本章介绍了以下典型的用户交互任务：

- 捕获应用程序范围内的键盘输入
- 捕获特定显示对象的键盘输入
- 捕获应用程序范围内的鼠标动作
- 捕获特定显示对象的鼠标输入
- 创建拖放交互性
- 创建自定义鼠标光标（鼠标指针）
- 将新行为添加到上下文菜单中
- 管理焦点

重要概念和术语

在继续阅读本章内容之前，一定要先熟悉以下重要的用户交互术语：

- 字符代码：表示当前字符集中的字符（与在键盘上所按的键关联）的数字代码。例如，尽管“D”和“d”是由美国英语键盘上的同一个键创建的，但它们具有不同的字符代码。
- 上下文菜单：当用户右键单击或使用特定键盘 - 鼠标组合时显示的菜单。上下文菜单命令通常直接应用于已单击的内容。例如，某个图像的上下文菜单可能包含用于在单独窗口中显示该图像以及下载该图像的命令。
- 焦点：指示选定元素是活动元素，并且它是键盘或鼠标交互的目标。
- 键控代码：对应于键盘上的实际键的数字代码。

完成本章中的示例

在学习本章的过程中，您可能想要自己动手测试一些范例代码清单。由于本章介绍的是如何在 ActionScript 中处理用户输入，因此本章中几乎所有的代码清单都涉及操作某一类型的显示对象 — 通常是文本字段或任何 **InteractiveObject** 子类。对于这些示例而言，显示对象可以是已经在 Adobe® Flash® CS4 Professional 中创建并放置在舞台上的显示对象，也可以是使用 ActionScript 创建的显示对象。测试范例涉及在 Flash Player 或 Adobe® AIR™ 中查看结果，并与范例交互以查看代码的效果。

要测试本章中的代码清单，请执行以下操作：

- 1 使用 Flash 创作工具创建空文档
- 2 在时间轴上选择一关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 在舞台上创建一个实例：
 - 如果代码引用一个文本字段，请使用“文本”工具在舞台上创建一个动态文本字段。
 - 否则，在舞台上创建一个按钮或影片剪辑元件实例。
- 5 选择该文本字段、按钮或影片剪辑，并在“属性”检查器中为它指定一个实例名。该名称应与范例代码中的显示对象的名称匹配 — 例如，如果代码操作一个名为 myDisplayObject 的对象，则您的舞台对象也要命名为 myDisplayObject。
- 6 通过“控制”>“测试影片”运行程序。

在屏幕上，将按照代码中的指定操作显示对象。

捕获键盘输入

从 **InteractiveObject** 类继承交互模型的显示对象可以使用事件侦听器来响应键盘事件。例如，您可以将事件侦听器放在舞台上以侦听并响应键盘输入。在以下代码中，事件侦听器捕获一个按键，并显示键名和键控代码属性：

```
function reportKeyDown(event:KeyboardEvent):void
{
    trace("Key Pressed: " + String.fromCharCode(event.charCodeAt) + " (character code: " + event.charCodeAt +
")");
}
stage.addEventListener(KeyboardEvent.KEY_DOWN, reportKeyDown);
```

有些键（如 Ctrl 键）虽然没有字型表示形式，但也能生成事件。

在上面的代码示例中，键盘事件侦听器捕获了整个舞台的键盘输入。也可以为舞台上的特定显示对象编写事件侦听器；当对象具有焦点时将触发该事件侦听器。

在以下示例中，仅当用户在 **TextField** 实例内键入内容时，才会在“输出”面板中反映键击。按下 Shift 键可暂时将 **TextField** 的边框颜色更改为红色。

此代码假定舞台上有一个名为 tf 的 **TextField** 实例。

```

tf.border = true;
tf.type = "input";
tf.addEventListener(KeyboardEvent.KEY_DOWN, reportKeyDown);
tf.addEventListener(KeyboardEvent.KEY_UP, reportKeyUp);

function reportKeyDown(event:KeyboardEvent):void
{
    trace("Key Pressed: " + String.fromCharCode(event.charCodeAt) + " (key code: " + event.keyCode + "
character code: " + event.charCodeAt + ")");
    if (event.keyCode == Keyboard.SHIFT) tf.borderColor = 0xFF0000;
}

function reportKeyUp(event:KeyboardEvent):void
{
    trace("Key Released: " + String.fromCharCode(event.charCodeAt) + " (key code: " + event.keyCode + "
character code: " + event.charCodeAt + ")");
    if (event.keyCode == Keyboard.SHIFT)
    {
        tf.borderColor = 0x000000;
    }
}

```

TextField 类还会报告 `textInput` 事件，当用户输入文本时，您可以倾听该事件。有关详细信息，请参阅第 365 页的“[捕获文本输入](#)”。

了解键控代码和字符代码

您可以访问键盘事件的 `keyCode` 和 `charCode` 属性，以确定按下了哪个键，然后触发其它动作。`keyCode` 属性为数值，与键盘上的某个键的值相对应。`charCode` 属性是该键在当前字符集中的数值。（默认字符集是 UTF-8，它支持 ASCII。）

键控代码值与字符值之间的主要区别是：键控代码值表示键盘上的特定键（数字键盘上的 1 与最上面一排键中的 1 不同，但生成的“1”键与生成“!”的键是同一个键），字符值表示特定字符（字符 R 与 r 是不同的）。

注：有关各个键与其在 ASCII 中的字符代码值之间的映射，请参阅 ActionScript 语言参考中的 [flash.ui.Keyboard](#) 类。

键与其键控代码之间的映射取决于设备和操作系统。因此，不应使用键映射来触发动作，而应使用 `Keyboard` 类提供的预定义常量值来引用相应的 `keyCode` 属性。例如，不要使用 `Shift` 的键映射，而应使用 `Keyboard.SHIFT` 常量（如上面的代码范例中所示）。

了解 `KeyboardEvent` 的优先顺序

与其它事件一样，键盘事件序列是由显示对象层次结构决定的，而不是由在代码中分配 `addEventListener()` 方法的顺序决定的。

例如，假定您将名为 `tf` 的文本字段放在名为 `container` 的影片剪辑内，并在这两个实例中添加键盘事件的事件倾听器，如下例所示：

```

container.addEventListener(KeyboardEvent.KEY_DOWN, reportKeyDown);
container.tf.border = true;
container.tf.type = "input";
container.tf.addEventListener(KeyboardEvent.KEY_DOWN, reportKeyDown);

function reportKeyDown(event:KeyboardEvent):void
{
    trace(event.currentTarget.name + " hears key press: " + String.fromCharCode(event.charCodeAt) + " (key
code: " + event.keyCode + " character code: " + event.charCodeAt + ")");
}

```

由于文本字段及其父容器上都有倾听器，因此，将为 `TextField` 内的每次键击调用两次 `reportKeyDown()` 函数。请注意，对于每次按键操作，文本字段在 `container` 影片剪辑调度事件之前调度事件。

操作系统和 Web 浏览器先于 Adobe Flash Player 或 AIR 处理键盘事件。例如，在 Microsoft Internet Explorer 中按 Ctrl+W 将先关闭浏览器窗口，然后再由包含的任何 SWF 文件调度键盘事件。

捕获鼠标输入

鼠标单击将创建鼠标事件，这些事件可用来触发交互式功能。您可以将事件侦听器添加到舞台上以侦听在 SWF 文件中任何位置发生的鼠标事件。也可以将事件侦听器添加到舞台上从 `InteractiveObject` 进行继承的对象（例如，`Sprite` 或 `MovieClip`）中；单击该对象时将触发这些侦听器。

与键盘事件一样，鼠标事件也会冒泡。在下面的示例中，由于 `square` 是 `Stage` 的子级，因此，单击正方形时，将从 `Sprite square` 和 `Stage` 对象中调度该事件：

```
var square:Sprite = new Sprite();
square.graphics.beginFill(0xFF0000);
square.graphics.drawRect(0,0,100,100);
square.graphics.endFill();
square.addEventListener(MouseEvent.CLICK, reportClick);
square.x =
square.y = 50;
addChild(square);

stage.addEventListener(MouseEvent.CLICK, reportClick);

function reportClick(event:MouseEvent):void
{
    trace(event.currentTarget.toString() + " dispatches MouseEvent. Local coords [" + event.localX + "," + event.localY + "] Stage coords [" + event.stageX + "," + event.stageY + "]");
}
```

请注意，在上面的示例中，鼠标事件包含有关单击的位置信息。`localX` 和 `localY` 属性包含显示链中最低级别的子级上的单击位置。例如，单击 `square` 左上角时将报告本地坐标 [0,0]，因为它是 `square` 的注册点。或者，`stageX` 和 `stageY` 属性是指单击位置在舞台上的全局坐标。同一单击报告这些坐标为 [50,50]，因为 `square` 已移到这些坐标上。取决于响应用户交互的方式，这两种坐标对可能是非常有用的。

`MouseEvent` 对象还包含 `altKey`、`ctrlKey` 和 `shiftKey` 布尔属性。可以使用这些属性来检查在鼠标单击时是否还按下了 Alt、Ctrl 或 Shift 键。

创建拖放功能

通过使用拖放功能，用户可以在按鼠标左键的同时选择对象，将该对象移到屏幕上的新位置，然后松开鼠标左键以将其放在新位置上。下面的代码显示了一个这样的示例：

```
import flash.display.Sprite;
import flash.events.MouseEvent;

var circle:Sprite = new Sprite();
circle.graphics.beginFill(0xFFCC00);
circle.graphics.drawCircle(0, 0, 40);

var target1:Sprite = new Sprite();
target1.graphics.beginFill(0xCCFF00);
target1.graphics.drawRect(0, 0, 100, 100);
target1.name = "target1";

var target2:Sprite = new Sprite();
target2.graphics.beginFill(0xCCFF00);
target2.graphics.drawRect(0, 200, 100, 100);
target2.name = "target2";

addChild(target1);
addChild(target2);
addChild(circle);

circle.addEventListener(MouseEvent.MOUSE_DOWN, mouseDown)

function mouseDown(event:MouseEvent):void
{
    circle.startDrag();
}
circle.addEventListener(MouseEvent.MOUSE_UP, mouseReleased);

function mouseReleased(event:MouseEvent):void
{
    circle.stopDrag();
    trace(circle.dropTarget.name);
}
```

有关详细信息，请参阅第 241 页的“[改变位置](#)”中有关创建拖放交互的部分。

自定义鼠标光标

可以将鼠标光标（鼠标指针）隐藏或交换为舞台上的任何显示对象。要隐藏鼠标光标，请调用 `Mouse.hide()` 方法。可通过以下方式来自定义光标：调用 `Mouse.hide()`，侦听舞台上是否发生 `MouseEvent.MOUSE_MOVE` 事件，以及将显示对象（自定义光标）的坐标设置为事件的 `stageX` 和 `stageY` 属性。下面的示例说明了此任务的基本执行过程：

```
var cursor:Sprite = new Sprite();
cursor.graphics.beginFill(0x000000);
cursor.graphics.drawCircle(0,0,20);
cursor.graphics.endFill();

addChild(cursor);

stage.addEventListener(MouseEvent.MOUSE_MOVE, redrawCursor);
Mouse.hide();

function redrawCursor(event:MouseEvent):void
{
    cursor.x = event.stageX;
    cursor.y = event.stageY;
}
```

自定义上下文菜单

从 `InteractiveObject` 类进行继承的每个对象可以具有唯一的上下文菜单，用户在 SWF 文件内右键单击时将显示该菜单。默认情况下，菜单中包含几个命令，其中包括“前进”、“后退”、“打印”、“品质”和“缩放”。

除了“设置”和“关于”命令外，您可以从菜单中删除所有其它默认命令。如果将 `Stage` 属性 `showDefaultContextMenu` 设置为 `false`，则会从上下文菜单中删除这些命令。

要为特定显示对象创建自定义的上下文菜单，请创建 `ContextMenu` 类的一个新实例，调用 `hideBuiltInItems()` 方法，并将该实例分配给该 `DisplayObject` 实例的 `contextMenu` 属性。下面的示例为一个动态绘制的正方形提供了一个上下文菜单命令，用于将其更改为随机颜色：

```
var square:Sprite = new Sprite();
square.graphics.beginFill(0x000000);
square.graphics.drawRect(0,0,100,100);
square.graphics.endFill();
square.x =
square.y = 10;
addChild(square);

var menuItem:ContextMenuItem = new ContextMenuItem("Change Color");
menuItem.addEventListener(ContextMenuEvent.MENU_ITEM_SELECT,changeColor);
var customContextMenu:ContextMenu = new ContextMenu();
customContextMenu.hideBuiltInItems();
customContextMenu.customItems.push(menuItem);
square.contextMenu = customContextMenu;

function changeColor(event:ContextMenuEvent):void
{
    square.transform.colorTransform = getRandomColor();
}
function getRandomColor():ColorTransform
{
    return new ColorTransform(Math.random(), Math.random(), Math.random(),1,(Math.random() * 512) - 255,
(Math.random() * 512) -255, (Math.random() * 512) - 255, 0);
}
```

管理焦点

交互式对象可以按编程方式或通过用户动作来获得焦点。在这两种情况下，设置焦点会将对象的 `focus` 属性更改为 `true`。另外，如果将 `tabEnabled` 属性设置为 `true`，用户可通过按 Tab 将焦点从一个对象传递到另一个对象。请注意，默认情况下，`tabEnabled` 值为 `false`，但以下情况除外：

- 对于 `SimpleButton` 对象，值为 `true`。
- 对于输入文本字段，该值为 `true`。
- 对于 `buttonMode` 设置为 `true` 的 `Sprite` 或 `MovieClip` 对象，该值为 `true`。

在上述各种情况下，都可以为 `FocusEvent.FOCUS_IN` 或 `FocusEvent.FOCUS_OUT` 添加侦听器，以便在焦点更改时提供其它行为。这对文本字段和表单尤其有用，但也可以用于 `sprite`、影片剪辑或从 `InteractiveObject` 类进行继承的任何对象。下面的示例说明了如何使用 Tab 键启用焦点循环切换，以及如何响应后续的焦点事件。在本例中，每个正方形在收到焦点时将改变颜色。

注：Flash 创作工具使用键盘快捷键来管理焦点；因此，若要正确模拟焦点管理，应在浏览器或 AIR 中测试 SWF 文件，而不是在 Flash 中进行测试。

```
var rows:uint = 10;
var cols:uint = 10;
var rowSpacing:uint = 25;
var colSpacing:uint = 25;
var i:uint;
var j:uint;
for (i = 0; i < rows; i++)
{
    for (j = 0; j < cols; j++)
    {
        createSquare(j * colSpacing, i * rowSpacing, (i * cols) + j);
    }
}

function createSquare(startX:Number, startY:Number, tabNumber:uint):void
{
    var square:Sprite = new Sprite();
    square.graphics.beginFill(0x000000);
    square.graphics.drawRect(0, 0, colSpacing, rowSpacing);
    square.graphics.endFill();
    square.x = startX;
    square.y = startY;
    square.tabEnabled = true;
    squareTabIndex = tabNumber;
    square.addEventListener(FocusEvent.FOCUS_IN, changeColor);
    addChild(square);
}
function changeColor(event:FocusEvent):void
{
    event.target.transform.colorTransform = getRandomColor();
}
function getRandomColor():ColorTransform
{
    // Generate random values for the red, green, and blue color channels.
    var red:Number = (Math.random() * 512) - 255;
    var green:Number = (Math.random() * 512) - 255;
    var blue:Number = (Math.random() * 512) - 255;

    // Create and return a ColorTransform object with the random colors.
    return new ColorTransform(1, 1, 1, 1, red, green, blue, 0);
}
```

示例：WordSearch

此示例通过处理鼠标事件来说明用户交互。用户在一个由随机字母组成的网格上构造尽可能多的词，拼写方法是：在网格中进行水平或垂直移动，但同一个字母只允许使用一次。此示例演示了 ActionScript 3.0 的下列功能：

- 动态构造组件网格
- 响应鼠标事件
- 根据用户交互维护分数

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可以在 Samples/WordSearch 文件夹中找到 WordSearch 应用程序文件。该应用程序包含以下文件：

文件	描述
WordSearch.as	此类提供了应用程序的主要功能。
WordSearch.fla 或 WordSearch.mxml	Flex 或 Flash 的主应用程序文件（分别为 MXML 和 FLA 格式）。
dictionary.txt	此文件用于确定拼写的词能否得分以及拼写是否正确。

加载字典

要创建一个涉及查找词的游戏，您需要使用字典。本示例包含一个名为 `dictionary.txt` 的文本文件，该文件包含以回车符分隔的词列表。创建名为 `words` 的数组后，`loadDictionary()` 函数将请求加载此文件，成功加载后，此文件将变成一个很长的字符串。通过使用 `split()` 方法，在回车符（字符代码 10）或换行符（字符代码 13）的每个实例处断开，可以将该字符串分析为词的数组。这种分析是在 `dictionaryLoaded()` 函数中实现的：

```
words = dictionaryText.split(String.fromCharCode(13, 10));
```

创建用户界面

在存储这些词后，您就可以设置用户界面了。请创建两个 `Button` 实例：一个用于提交词，另一个用于清除当前拼写的词。在每种情况下，您必须通过侦听该按钮所广播的 `MouseEvent.CLICK` 事件并随后调用一个函数来响应用户输入。在 `setupUI()` 函数中，此代码在两个按钮上创建侦听器：

```
submitWordButton.addEventListener(MouseEvent.CLICK, submitWord);  
clearWordButton.addEventListener(MouseEvent.CLICK, clearWord);
```

生成游戏板

游戏板是由随机字母组成的网格。在 `generateBoard()` 函数中，二维网格是通过将一个循环嵌套到另一个循环中创建的。第一个循环增加行数，第二个循环增加每行的总列数。由这些行和列创建的每个单元格包含一个按钮，它表示游戏板上的一个字母。

```

private function generateBoard(startX:Number, startY:Number, totalRows:Number, totalCols:Number,
buttonSize:Number):void
{
    buttons = new Array();
    var colCounter:uint;
    var rowCounter:uint;
    for (rowCounter = 0; rowCounter < totalRows; rowCounter++)
    {
        for (colCounter = 0; colCounter < totalCols; colCounter++)
        {
            var b:Button = new Button();
            b.x = startX + (colCounter*buttonSize);
            b.y = startY + (rowCounter*buttonSize);
            b.addEventListener(MouseEvent.CLICK, letterClicked);
            b.label = getRandomLetter().toUpperCase();
            b.setSize(buttonSize,buttonSize);
            b.name = "buttonRow"+rowCounter+"Col"+colCounter;
            addChild(b);

            buttons.push(b);
        }
    }
}

```

虽然仅在一行中添加了 `MouseEvent.CLICK` 事件的侦听器，但由于该侦听器位于 `for` 循环中，因此会将其分配给每个 `Button` 实例。此外，还会为每个按钮分配一个从其行和列位置派生的名称，这为以后在代码中引用每个按钮的行和列提供了一种简便的方法。

通过用户输入来构造词

拼词规则是：选择垂直或水平相邻的字母，但每个字母只允许使用一次。每次单击时将生成一个鼠标事件，此时必须检查用户正在拼写的词，以确保该词正确地从以前单击的字母继续进行拼写。否则，将删除以前的词并开始拼写一个新词。此检查在 `isLegalContinuation()` 方法中执行。

```

private function isLegalContinuation(prevButton:Button, currButton:Button):Boolean
{
    var currButtonRow:Number = Number(currButton.name.charAt(currButton.name.indexOf("Row") + 3));
    var currButtonCol:Number = Number(currButton.name.charAt(currButton.name.indexOf("Col") + 3));
    var prevButtonRow:Number = Number(prevButton.name.charAt(prevButton.name.indexOf("Row") + 3));
    var prevButtonCol:Number = Number(prevButton.name.charAt(prevButton.name.indexOf("Col") + 3));

    return ((prevButtonCol == currButtonCol && Math.abs(prevButtonRow - currButtonRow) <= 1) ||
           (prevButtonRow == currButtonRow && Math.abs(prevButtonCol - currButtonCol) <= 1));
}

```

`String` 类的 `charAt()` 和 `indexOf()` 方法从当前单击的按钮和上次单击的按钮中检索相应的行和列。如果行或列未改变，或者行或列已改变但与上次单击位置的行或列之间相差不超过一行或一列，`isLegalContinuation()` 方法将返回 `true`。如果要更改游戏规则并允许斜向拼写，您可以删除对未改变的行或列的检查，最后一行将如下所示：

```
return (Math.abs(prevButtonRow - currButtonRow) <= 1) && Math.abs(prevButtonCol - currButtonCol) <= 1);
```

检查词的提交

要完成游戏的代码，您需要提供检查词的提交和计算分数的机制。`searchForWord()` 方法包含这两项功能：

```
private function searchForWord(str:String):Number
{
    if (words && str)
    {
        var i:uint = 0
        for (i = 0; i < words.length; i++)
        {
            var thisWord:String = words[i];
            if (str == words[i])
            {
                return i;
            }
        }
        return -1;
    }
    else
    {
        trace("WARNING: cannot find words, or string supplied is null");
    }
    return -1;
}
```

此函数循环访问字典中所有的词。如果用户的词与字典中的词匹配，则返回该词在字典中的位置。如果该位置有效，submitWord() 方法随后将检查响应并更新分数。

自定义

类的开头是几个常量。可通过修改这些变量来修改此游戏。例如，您可以通过增大 TOTAL_TIME 变量的值来更改可用于玩该游戏的时间。还可以稍微增大 PERCENT_VOWELS 变量的值来增大找到词的可能性。

第 27 章：网络与通信

本章介绍如何使 SWF 文件能与外部文件和其它 Adobe Flash Player 及 Adobe AIR 实例进行通信。还介绍如何从外部源加载数据，在 Java 服务器和 Flash Player 之间发送消息，以及使用 `FileReference` 和 `FileReferenceList` 类执行文件上载和下载。

网络和通信基础知识

网络和通信简介

当构建更复杂的 ActionScript 应用程序时，通常需要与服务器端脚本进行通信，或者从外部 XML 文件或文本文件加载数据。`flash.net` 包中包含用于通过 Internet 发送和接收数据（例如，从远程 URL 加载内容、与其它 Flash Player 或 AIR 实例进行通信以及连接到远程网站）的类。

而在 ActionScript 3.0 中，可以使用 `URLLoader` 和 `URLRequest` 类加载外部文件。可随后使用特定类来访问数据，具体取决于加载的数据类型。例如，如果将远程内容的格式设置为名称 - 值对，则可以使用 `URLVariables` 类来分析服务器结果。或者，如果使用 `URLLoader` 和 `URLRequest` 类加载的文件是远程 XML 文档，则可以使用 `XML` 类的构造函数、`XMLDocument` 类的构造函数或 `XMLDocument.parseXML()` 方法来分析 XML 文档。这样，您便可以简化 ActionScript 代码，因为无论是使用 `URLVariables`、`XML` 还是某个其它类来分析和处理远程数据，用于加载外部文件的代码都是相同的。

`flash.net` 包中还包含用于其它类型的远程通信的类。这些类包括 `FileReference` 类（用于将文件上载到服务器以及从服务器下载文件）、`Socket` 和 `XMLSocket` 类（用于通过套接字连接直接与远程计算机进行通信）以及 `NetConnection` 和 `NetStream` 类（用于与特定于 Flash 的服务器资源（如 Flash Media Server 和 Flash Remoting 服务器）进行通信以及加载视频文件）。

最后，`flash.net` 包中包含用于用户本地计算机上的通信的类。这些类包括 `LocalConnection` 类（用于在一台计算机上运行的两个或多个 SWF 文件之间进行通信）和 `SharedObject` 类（用于将数据存储在用户的计算机上，并在以后返回到应用程序时检索这些数据）。

常见网络和通信用任务

下面的列表说明了需要从 ActionScript 中执行的、与外部通信有关的常见任务；本章中对这些任务进行了介绍：

- 从外部文件或服务器脚本中加载数据
- 将数据发送到服务器脚本
- 与其它本地 SWF 文件通信
- 处理二进制套接字连接
- 与 XML 套接字通信
- 存储永久本地数据
- 将文件上载到服务器
- 将文件从服务器下载到用户计算机

重要概念和术语

以下参考列表包含将会在本章中遇到的重要术语：

- 外部数据：此类数据以某些形式存储在 SWF 文件外部，并在需要时加载到 SWF 文件中。可以将此数据存储在直接加载的文件中，存储在数据库中，或者采用可通过调用服务器上所运行的脚本或程序进行检索其它形式存储。

- URL 编码变量：URL 编码格式提供了一种在单个文本字符串中表示多个变量（变量名和值对）的方法。各个变量采用 `name=value` 格式。各个变量（即各个名称 - 值对）之间用 & 符隔开，如下所示：`variable1=value1&variable2=value2`。这样，便可以将不限数量的变量作为一条消息进行发送。
- MIME 类型：用于在 Internet 通信中标识给定文件类型的标准代码。任何给定文件类型都具有用于对其进行标识的特定代码。发送文件或消息时，计算机（如 Web 服务器或用户的 Flash Player 或 AIR 实例）将指定要发送的文件类型。
- HTTP：超文本传输协议，这是一种标准格式，用于传送通过 Internet 发送的网页和其它各种类型的内容。
- 请求方法：当程序（如 Flash Player 或 Web 浏览器）将消息（称为 HTTP 请求）发送到 Web 服务器时，可以使用以下两种方法之一将发送的任何数据嵌入到请求中；这两种方法是两个“请求方法”，即 GET 和 POST。在服务器端，接收请求的程序需要查看相应的请求部分以查找数据，因此，用于从 ActionScript 发送数据的请求方法应与用于在服务器上读取该数据的请求方法相匹配。
- 套接字连接：用于两台计算机之间的通信的永久连接。
- 上载：将文件发送到另一台计算机。
- 下载：从另一台计算机检索文件。

使用 IPv6 地址

Flash Player 9.0.115.0 及更高版本支持 IPv6（Internet 协议版本 6）。IPv6 是支持 128 位地址的 Internet 协议版本（它是支持 32 位地址的早期 IPv4 协议的改进版本）。您可能需要在网络接口中激活 IPv6。有关详细信息，请参阅承载数据的操作系统的帮助。

如果承载系统中支持 IPv6，您可以在用方括号（[]）括起的 URL 中指定数字 IPv6 文本地址，如下所示：

```
rtmp://[2001:db8:ccc3:ffff:0:444d:555e:666f]:1935/test
```

Flash Player 根据以下规则返回 IPv6 字面值：

- Flash Player 返回长形式的 IPv6 地址字符串。
- IP 值没有双冒号缩写词。
- 十六进制数字全小写。
- IPv6 地址包含在方括号（[]）中。
- 每个四重地址都输出为 0 到 4 个十六进制数字（省略前导零）。
- 内容全为零的四重地址输出为单个零（而不是双冒号），下表所列例外情况除外。

Flash Player 返回的 IPv6 值具有以下例外：

- 未指定的 IPv6 地址（内容全为零）输出为 [::]。
- 环回或本地主机 IPv6 地址输出为 [::1]。
- IPv4 映射（转换为 IPv6）地址输出为 [::ffff:a.b.c.d]，其中 a.b.c.d 为典型的 IPv4 点分十进制值。
- IPv4 兼容地址输出为 [::a.b.c.d]，其中 a.b.c.d 为典型的 IPv4 点分十进制值。

完成本章中的示例

学习本章的过程中，您可能想要测试示例代码清单。本章中有几个代码清单加载外部数据或执行某些其它类型的通信；这些范例通常包括 `trace()` 函数调用，因此会在“输出”面板中显示示例的运行结果。其它示例则实际执行某一功能，例如将文件上传到服务器。测试这些示例将涉及与 SWF 交互并确认它们执行了所宣称的操作。

这些代码示例分为两类。一些示例列表是在假定代码在独立脚本（例如，附加到 Flash 文档中的关键帧的脚本）中的情况下编写的。要测试这些示例，请执行以下操作：

- 1 创建一个新的 Flash 文档。

- 2 选择时间轴的第 1 帧中的关键帧，并打开“动作”面板。
- 3 将代码清单复制到“脚本”窗格中。
- 4 从主菜单中，选择“控制”>“测试影片”以创建 SWF 文件并测试该示例。

其它示例代码清单编写为类的形式；示例类的预期功能是充当 Flash 文档的文档类。要测试这些示例，请执行以下操作：

- 1 创建一个空的 Flash 文档并将它保存到您的计算机上。
- 2 创建一个新的 ActionScript 文件，并将它保存到 Flash 文档所在的目录中。文件名应与代码清单中的类的名称一致。例如，如果代码清单定义一个名为“UploadTest”的类，则将 ActionScript 文件保存为“UploadTest.as”。
- 3 将代码清单复制到 ActionScript 文件中并保存该文件。
- 4 在 Flash 文档中，单击舞台或夹纸板的空白部分，以激活文档的“属性”检查器。
- 5 在“属性”检查器的“文档类”字段中，输入您从文本中复制的 ActionScript 类的名称。
- 6 使用“控制”>“测试影片”运行程序并测试该示例。

最后，本章中的一些示例涉及与在服务器上运行的程序进行交互。这些示例包括可用来创建测试示例所需的服务器程序的代码；您将需要在 Web 服务器计算机上设置适当的应用程序来测试这些示例。

处理外部数据

ActionScript 3.0 包含用于从外部源加载数据的机制。这些源可以是静态内容（如文本文件），也可以是动态内容（如从数据库检索数据的 Web 脚本）。可以使用多种不同的方法来设置数据的格式，并且 ActionScript 提供了用于解码和访问数据的功能。也可以在检索数据的过程中将数据发送到外部服务器。

使用 URLLoader 类和 URLVariables 类

ActionScript 3.0 使用 URLLoader 和 URLVariables 类来加载外部数据。URLLoader 类以文本、二进制数据或 URL 编码变量的形式从 URL 下载数据。URLLoader 类用于下载文本文件、XML 或其它要用于数据驱动的动态 ActionScript 应用程序中的信息。URLLoader 类使用 ActionScript 3.0 高级事件处理模型，使用该模型可以侦听诸如 complete、httpStatus、ioError、open、progress 和 securityError 等事件。新事件处理模型在 ActionScript 2.0 的基础上大大改进了对 LoadVars.onData、LoadVars.onHTTPStatus 和 LoadVars.onLoad 事件处理函数的支持，可以更高效地处理错误和事件。有关处理事件的详细信息，请参阅第 207 页的“[处理事件](#)”

与早期版本 ActionScript 中的 XML 和 LoadVars 类非常相似，URLLoader URL 的数据在下载完成之前不可用。尽管如果文件加载速度太快，可能不会调度 ProgressEvent.PROGRESS 事件，但您可以通过侦听要调度的 flash.events.ProgressEvent.PROGRESS 事件来监视下载进度（已加载的字节数和总字节数）。成功下载文件后，将调度 flash.events.Event.COMPLETE 事件。加载的数据将从 UTF-8 或 UTF-16 编码被解码为字符串。

注：如果没有为 URLRequest.contentType 设置值，则以 application/x-www-form-urlencoded 的形式发送值。

URLLoader.load() 方法（以及 URLLoader 类的构造函数，可选）使用一个参数，即 request，该参数是一个 URLRequest 实例。URLRequest 实例包含单个 HTTP 请求的所有信息，如目标 URL、请求方法（GET 或 POST）、附加标头信息以及 MIME 类型（例如，当上载 XML 内容时）。

例如，要将 XML 数据包上载到服务器端脚本，您可以使用下面的 ActionScript 3.0 代码：

```
var secondsUTC:Number = new Date().time;
var dataXML:XML =
<login>
<time>{secondsUTC}</time>
<username>Ernie</username>
<password>guru</password>
</login>;
var request:URLRequest = new URLRequest("http://www.yourdomain.com/login.cfm");
request.contentType = "text/xml";
request.data = dataXML.toXMLString();
request.method = URLRequestMethod.POST;
var loader:URLLoader = new URLLoader();
try
{
    loader.load(request);
}
catch (error:ArgumentError)
{
    trace("An ArgumentError has occurred.");
}
catch (error:SecurityError)
{
    trace("A SecurityError has occurred.");
}
```

上面的代码片断创建了一个名为 `dataXML` 的 XML 实例，该实例包含要发送到服务器的 XML 数据包。接下来，将 `URLRequest` `contentType` 属性设置为 "text/xml"，将 `URLRequest` `data` 属性设置为 XML 数据包的内容（通过 `XML.toXMLString()` 方法将该内容转换为字符串）。最后，创建一个新的 `URLLoader` 实例，并使用 `URLLoader.load()` 方法将请求发送到远程脚本。

可以使用三种方式指定要在 URL 请求中传递的参数：

- 在 `URLVariables` 构造函数中
- 在 `URLVariables.decode()` 方法中
- 作为 `URLVariables` 对象本身中的特定属性

当定义 `URLVariables` 构造函数或 `URLVariables.decode()` 方法中的变量时，需要确保对 "and" 符进行 URL 编码，因为它具有特殊含义并作为分隔符使用。例如，由于 "and" 符作为参数的分隔符使用，因此当传递 "and" 符时，需要将其从 & 更改为 %26 来对 "and" 符进行 URL 编码。

从外部文档加载数据

当使用 ActionScript 3.0 生成动态应用程序时，最好从外部文件或服务器端脚本加载数据。这样，您不必编辑或重新编译 ActionScript 文件，即可生成动态应用程序。例如，如果生成“每日提示”应用程序，则可以编写一个服务器端脚本，该脚本每天从数据库检索一次随机提示并将其保存到文本文件。然后，ActionScript 应用程序可以加载静态文本文件的内容，而不必每次查询数据库。

下面的片断创建 `URLRequest` 和 `URLLoader` 对象，用于加载外部文本文件 `params.txt` 的内容：

```
var request:URLRequest = new URLRequest("params.txt");
var loader:URLLoader = new URLLoader();
loader.load(request);
```

可以将上面的片断简化如下：

```
var loader:URLLoader = new URLLoader(new URLRequest("params.txt"));
```

默认情况下，如果您未定义请求方法，则 Flash Player 和 Adobe AIR 会使用 HTTP GET 方法加载内容。如果要使用 POST 方法发送数据，则需要使用静态常量 `URLRequestMethod.POST` 将 `request.method` 属性设置为 POST，如以下代码所示：

```
var request:URLRequest = new URLRequest("sendfeedback.cfm");
request.method = URLRequestMethod.POST;
```

在运行时加载的外部文档 `params.txt` 包含以下数据：

```
monthNames=January,February,March,April,May,June,July,August,September,October,November,December&dayNames
=Sunday,Monday,Tuesday,Wednesday,Thursday,Friday,Saturday
```

该文件包含两个参数，即 `monthNames` 和 `dayNames`。每个参数包含一个逗号分隔列表，该列表被分解为字符串。可以使用 `String.split()` 方法将此列表拆分为数组。

 不要将保留字或语言构造作为外部数据文件中的变量名称，因为这样做会使代码的读取和调试变得更困难。

加载数据后，将调度 `Event.COMPLETE` 事件，随后将可以在 `URLLoader` 的 `data` 属性中使用外部文档的内容，如以下代码所示：

```
private function completeHandler(event:Event):void
{
    var loader2:URLLoader = URLLoader(event.target);
    trace(loader2.data);
}
```

如果远程文档包含名称 - 值对，您可以通过传入加载文件的内容，使用 `URLVariables` 类来分析数据，如下所示：

```
private function completeHandler(event:Event):void
{
    var loader2:URLLoader = URLLoader(event.target);
    var variables:URLVariables = new URLVariables(loader2.data);
    trace(variables.dayNames);
}
```

外部文件中的每个名称 - 值对都创建为 `URLVariables` 对象中的一个属性。在上面的代码范例中，变量对象中的每个属性都被视为字符串。如果名称 - 值对的值是一个项目列表，则可以通过调用 `String.split()` 方法将字符串转换为数组，如下所示：

```
var dayNameArray:Array = variables.dayNames.split(",");
```

 如果从外部文本文件加载数值数据，则需要使用顶级函数（如 `int()`、`uint()` 或 `Number()`）将这些值转换为数值。

无需将远程文件的内容作为字符串加载和新建 `URLVariables` 对象，您可以将 `URLLoader.dataFormat` 属性设置为在 `URLLoaderDataFormat` 类中找到的静态属性之一。`URLLoader.dataFormat` 属性的三个可能值如下：

- `URLLoaderDataFormat.BINARY` — `URLLoader.data` 属性包含 `ByteArray` 对象中存储的二进制数据。
- `URLLoaderDataFormat.TEXT` — `URLLoader.data` 属性包含 `String` 对象中的文本。
- `URLLoaderDataFormat.VARIABLES` — `URLLoader.data` 属性包含 `URLVariables` 对象中存储的 URL 编码的变量。

下面的代码演示了如何通过将 `URLLoader.dataFormat` 属性设置为 `URLLoaderDataFormat.VARIABLES`，从而自动将加载的数据分析为 `URLVariables` 对象：

```
package
{
    import flash.display.Sprite;
    import flash.events.*;
    import flash.net.URLLoader;
    import flash.net.URLLoaderDataFormat;
    import flash.net.URLRequest;

    public class URLLoaderDataFormatExample extends Sprite
    {
        public function URLLoaderDataFormatExample()
        {
            var request:URLRequest = new URLRequest("http://www.[yourdomain].com/params.txt");
            var variables:URLLoader = new URLLoader();
            variables.dataFormat = URLLoaderDataFormat.VARIABLES;
            variables.addEventListener(Event.COMPLETE, completeHandler);
            try
            {
                variables.load(request);
            }
            catch (error:Error)
            {
                trace("Unable to load URL: " + error);
            }
        }
        private function completeHandler(event:Event):void
        {
            var loader:URLLoader = URLLoader(event.target);
            trace(loader.data.dayNames);
        }
    }
}
```

注: URLLoader.dataFormat 的默认值为 URLLoaderDataFormat.TEXT。

如下面的实例所示, 从外部文件加载 XML 与加载 URLVariables 相同。可以创建 URLRequest 和 URLLoader 实例, 然后使用它们下载远程 XML 文档。文件完全下载后, 调度 Event.COMPLETE 事件, 并将外部文件的内容转换为可使用 XML 方法和属性分析的 XML 实例。

```
package
{
    import flash.display.Sprite;
    import flash.errors.*;
    import flash.events.*;
    import flash.net.URLLoader;
    import flash.net.URLRequest;

    public class ExternalDocs extends Sprite
    {
        public function ExternalDocs()
        {
            var request:URLRequest = new URLRequest("http://www.[yourdomain].com/data.xml");
            var loader:URLLoader = new URLLoader();
            loader.addEventListener(Event.COMPLETE, completeHandler);
            try
            {
                loader.load(request);
            }
            catch (error:ArgumentError)
            {
                trace("An ArgumentError has occurred.");
            }
            catch (error:SecurityError)
            {
                trace("A SecurityError has occurred.");
            }
        }

        private function completeHandler(event:Event):void
        {
            var dataXML:XML = XML(event.target.data);
            trace(dataXML.toXMLString());
        }
    }
}
```

与外部脚本进行通信

除了加载外部数据文件，还可以使用 **URLVariables** 类将变量发送到服务器端脚本并处理服务器的响应。这是非常有用的，例如，如果您正在编写游戏，想要将用户的得分发送到服务器以计算是否应添加到高分列表中，甚至想要将用户的登录信息发送到服务器以进行验证。服务器端脚本可以处理用户名和密码，向数据库验证用户名和密码，然后返回用户提供的凭据是否有效的确认。

下面的片断创建一个名为 **variables** 的 **URLVariables** 对象，该对象创建称为 **name** 的新变量。接下来，创建一个 **URLRequest** 对象，该对象指定变量要发送到的服务器端脚本的 URL。然后，设置 **URLRequest** 对象的 **method** 属性，以便将变量作为 HTTP POST 请求发送。为了将 **URLVariables** 对象添加到 URL 请求，需要将 **URLRequest** 对象的 **data** 属性设置为早先创建的 **URLVariables** 对象。最后，创建 **URLLoader** 实例并调用 **URLLoader.load()** 方法，此方法用于启动该请求。

```
var variables:URLVariables = new URLVariables("name=Franklin");
var request:URLRequest = new URLRequest();
request.url = "http://www.[yourdomain].com/greeting.cfm";
request.method = URLRequestMethod.POST;
request.data = variables;
var loader:URLLoader = new URLLoader();
loader.dataFormat = URLLoaderDataFormat.VARIABLES;
loader.addEventListener(Event.COMPLETE, completeHandler);
try
{
    loader.load(request);
}
catch (error:Error)
{
    trace("Unable to load URL");
}

function completeHandler(event:Event):void
{
    trace(event.target.data.welcomeMessage);
}
```

下面的代码包含上面的示例中使用的 Adobe ColdFusion® greeting.cfm 文档的内容：

```
<cfif NOT IsDefined("Form.name") OR Len(Trim(Form.Name)) EQ 0>
    <cfset Form.Name = "Stranger" />
</cfif>
<cfoutput>welcomeMessage=#UrlEncodedFormat("Welcome, " & Form.name)#
</cfoutput>
```

连接到其它 Flash Player 和 AIR 实例

通过使用 LocalConnection 类，可以在不同的 Flash Player 和 AIR 实例（如 HTML 容器、嵌入或独立播放器中的 SWF）之间进行通信。这样，您便能构建可以在 Flash Player 和 AIR 实例之间共享数据（例如在 Web 浏览器中运行或嵌入在桌面应用程序中的 SWF 文件）的各种不同用途的应用程序。

LocalConnection 类

LocalConnection 类用于开发 SWF 文件，这些文件无需使用 fscommand() 方法或 JavaScript 即可向其它 SWF 文件发送指令。LocalConnection 对象只能在同一客户端计算机上运行的 SWF 文件间进行通信，但是它们可以在不同的应用程序中运行。例如，虽然放映文件维护本地信息，而基于浏览器的 SWF 进行的是远程连接，但在浏览器中运行的 SWF 文件和在放映文件中运行的 SWF 文件可以共享信息。（放映文件是以可作为独立应用程序运行的格式保存的 SWF 文件 — 即，放映文件嵌入在可执行文件中，因此不需要安装 Flash Player。）

可以使用 LocalConnection 对象在使用不同 ActionScript 版本的 SWF 之间进行通信：

- ActionScript 3.0 LocalConnection 对象可以与使用 ActionScript 1.0 或 2.0 创建的 LocalConnection 对象进行通信。
- ActionScript 1.0 或 2.0 LocalConnection 对象可以与使用 ActionScript 3.0 创建的 LocalConnection 对象进行通信。

Flash Player 可自动处理不同版本 LocalConnection 对象间的通信。

最简便的 LocalConnection 对象使用方法是只允许位于同一个域中的 LocalConnection 对象之间进行通信。这样，您就不必担心安全方面的问题。但如果您需要在不同域之间进行通讯，则有多种方法来实现安全性措施。有关详细信息，请参阅《ActionScript 3.0 语言和组件参考》中有关 send() 方法的 connectionName 参数的讨论以及 LocalConnection 类列表中的 allowDomain() 和 domain 条目。

 可以使用 LocalConnection 对象在一个 SWF 文件中收发数据，但是 Adobe 不建议这样做。相反，您应使用共享对象。

可以使用三种方式将回调方法添加到 LocalConnection 对象中：

- 使 LocalConnection 类成为子类，并添加方法。
- 将 LocalConnection.client 属性设置为实现方法的对象。
- 创建扩展 LocalConnection 的动态类，并动态附加方法。

添加回调方法的第一种方式是扩展 LocalConnection 类。您在自定义类中定义方法，而不是将它们动态添加到 LocalConnection 实例中。下面的代码说明了此方式：

```
package
{
    import flash.net.LocalConnection;
    public class CustomLocalConnection extends LocalConnection
    {
        public function CustomLocalConnection(connectionName:String)
        {
            try
            {
                connect(connectionName);
            }
            catch (error:ArgumentError)
            {
                // server already created/connected
            }
        }
        public function onMethod(timeString:String):void
        {
            trace("onMethod called at: " + timeString);
        }
    }
}
```

要新建 DynamicLocalConnection 类的实例，可以使用下面的代码：

```
var serverLC:CustomLocalConnection;
serverLC = new CustomLocalConnection("serverName");
```

添加回调方法的第二种方式是使用 LocalConnection.client 属性。这包括创建自定义类和将新实例分配给 client 属性，如下面的代码所示：

```
var lc:LocalConnection = new LocalConnection();
lc.client = new CustomClient();
```

LocalConnection.client 属性指示应调用的对象回调方法。在上面的代码中，client 属性设置为自定义类 CustomClient 的新实例。client 属性的默认值是当前 LocalConnection 实例。如果有两个具有同一方法集但是行为不同的数据处理函数，则可以使用 client 属性 — 例如，在某个应用程序中，一个窗口中的按钮切换另一个窗口中的视图。

要创建 CustomClient 类，可以使用下面的代码：

```
package
{
    public class CustomClient extends Object
    {
        public function onMethod(timeString:String):void
        {
            trace("onMethod called at: " + timeString);
        }
    }
}
```

添加回调方法的第三种方式是创建动态类并动态附加该方法，这与在早期版本的 ActionScript 中使用 LocalConnection 类非常相似，如下面的代码所示：

```
import flash.net.LocalConnection;
dynamic class DynamicLocalConnection extends LocalConnection {}
```

通过使用下面的代码，可以将回调方法动态添加到此类中：

```
var connection:DynamicLocalConnection = new DynamicLocalConnection();
connection.onMethod = this.onMethod;
// Add your code here.
public function onMethod(timeString:String):void
{
    trace("onMethod called at: " + timeString);
}
```

不建议使用上面这种添加回调方法的方式，因为该代码不是非常易于移植。另外，使用此方法创建本地连接会导致性能问题，因为访问动态属性比访问密封属性慢得多。

在两个 Flash Player 实例之间发送消息

可以使用 LocalConnection 类在 Flash Player 和 Adobe AIR 的不同实例之间进行通信。例如，可以在网页上有多个 Flash Player 实例，或者让 Flash Player 实例从弹出窗口中的 Flash Player 实例检索数据。

下面的代码定义一个本地连接对象，该对象作为服务器使用，接受来自其它 Flash Player 实例的传入调用：

```
package
{
    import flash.net.LocalConnection;
    import flash.display.Sprite;
    public class ServerLC extends Sprite
    {
        public function ServerLC()
        {
            var lc:LocalConnection = new LocalConnection();
            lc.client = new CustomClient1();
            try
            {
                lc.connect("conn1");
            }
            catch (error:Error)
            {
                trace("error:: already connected");
            }
        }
    }
}
```

此代码先创建一个名为 lc 的 LocalConnection 对象，然后将 client 属性设置为自定义类 CustomClient1。当另一个 Flash Player 实例调用此本地连接实例中的某方法时，Flash Player 在 CustomClient1 类中查找该方法。

当 Flash Player 实例连接到此 SWF 文件并尝试调用指定本地连接的任何方法时，系统会将请求发送到 client 属性指定的类（该属性被设置为 CustomClient1 类）：

```

package
{
    import flash.events.*;
    import flash.system.fscommand;
    import flash.utils.Timer;
    public class CustomClient1 extends Object
    {
        public function doMessage(value:String = ""):void
        {
            trace(value);
        }
        public function doQuit():void
        {
            trace("quitting in 5 seconds");
            this.close();
            var quitTimer:Timer = new Timer(5000, 1);
            quitTimer.addEventListener(TimerEvent.TIMER, closeHandler);
        }
        public function closeHandler(event:TimerEvent):void
        {
            fscommand("quit");
        }
    }
}

```

若要创建 LocalConnection 服务器, 请调用 LocalConnection.connect() 方法并提供唯一的连接名称。如果已存在具有指定名称的连接, 则会生成 ArgumentError 错误, 指出由于已经连接了该对象, 连接尝试失败。

下面的片断说明如何使用名称 conn1 创建新的套接字连接:

```

try
{
    connection.connect ("conn1");
}
catch (error:ArgumentError)
{
    trace("Error! Server already exists\n");
}

```

从辅助 SWF 文件连接到主 SWF 文件需要在发送方 LocalConnection 对象中新建 LocalConnection 对象, 然后使用连接名称和要执行的方法名称来调用 LocalConnection.send() 方法。例如, 要连接到早先创建的 LocalConnection 对象, 可以使用下面的代码:

```
sendingConnection.send("conn1", "doQuit");
```

此代码使用连接名称 conn1 连接到现有 LocalConnection 对象, 并调用远程 SWF 文件中的 doQuit() 方法。如果想要将参数发送到远程 SWF 文件, 可以在 send() 方法中的方法名称后指定附加参数, 如下面的片断所示:

```
sendingConnection.send("conn1", "doMessage", "Hello world");
```

连接到不同域中的 SWF 文档

要只允许从特定域进行通信, 可以调用 LocalConnection 类的 allowDomain() 或 allowInsecureDomain() 方法, 并传递包含允许访问此 LocalConnection 对象的一个或多个域的列表。

在早期版本的 ActionScript 中, LocalConnection.allowDomain() 和 LocalConnection.allowInsecureDomain() 是必须由开发人员实现的、且必须返回布尔值的回调方法。在 ActionScript 3.0 中, LocalConnection.allowDomain() 和 LocalConnection.allowInsecureDomain() 都是内置方法, 开发人员可以像调用 Security.allowDomain() 和 Security.allowInsecureDomain() 那样调用这两个内置方法, 传递要允许的一个或多个域的名称。

可以向 LocalConnection.allowDomain() 和 LocalConnection.allowInsecureDomain() 方法传递两个特殊值: * 和 localhost。星号值 (*) 表示允许从所有域访问。字符串 localhost 允许从本地安装的 SWF 文件调用 SWF 文件。

Flash Player 8 对本地 SWF 文件引入了安全限制。可以访问 Internet 的 SWF 文件还不能访问本地文件系统。如果指定 localhost，则任何本地 SWF 文件都可以访问 SWF 文件。如果 LocalConnection.send() 方法试图从调用代码没有访问权限的安全沙箱与 SWF 文件进行通信，则会调度 securityError 事件 (SecurityErrorEvent.SECURITY_ERROR)。若要解决此错误，可以在接收方的 LocalConnection.allowDomain() 方法中指定调用方的域。

如果仅在同一个域中的 SWF 文件之间实现通信，可以指定一个不以下划线 (_) 开头且不指定域名的 connectionName 参数（例如 myDomain:connectionName）。在 LocalConnection.connect(connectionName) 命令中使用相同的字符串。

如果要实现不同域中的 SWF 文件之间的通信，可以指定一个以下划线开头的 connectionName 参数。指定下划线使具有接收方 LocalConnection 对象的 SWF 文件更易于在域之间移植。下面是两种可能的情形：

- 如果 connectionName 字符串不以下划线开头，则 Flash Player 会添加一个包含超级域名称和一个冒号的前缀（例如 myDomain:connectionName）。虽然这可以确保您的连接不会与其它域中具有同一名称的连接冲突，但任何发送方 LocalConnection 对象都必须指定此超级域（例如 myDomain:connectionName）。如果将具有接收方 LocalConnection 对象的 SWF 文件移动到另一个域中，Flash Player 会更改前缀，以反映新的超级域（例如 anotherDomain:connectionName）。必须手动编辑所有发送方 LocalConnection 对象，以指向新超级域。
- 如果 connectionName 字符串以下划线开头（例如 _connectionName），Flash Player 不会向该字符串添加前缀。这意味着接收方和发送方 LocalConnection 对象都将使用相同的 connectionName 字符串。如果接收方对象使用 LocalConnection.allowDomain() 来指定可以接受来自任何域的连接，则可以将具有接收方 LocalConnection 对象的 SWF 文件移动到另一个域，而无需更改任何发送方 LocalConnection 对象。

套接字连接

在 ActionScript 3.0 中，可以使用两种不同类型的套接字连接：XML 套接字连接和二进制套接字连接。使用 XML 套接字，可以连接到远程服务器并创建服务器连接，该连接在显式关闭之前一直保持打开。这样，无需不断打开新服务器连接，就可以在服务器与客户端之间交换 XML 数据。使用 XML 套接字服务器的另一个好处是用户不需要显式请求数据。您无需请求即可从服务器发送数据，并且可以将数据发送到连接到 XML 套接字服务器的每个客户端。

XML 套接字连接要求目标服务器上存在套接字策略文件。有关详细信息，请参阅第 592 页的“[网站控制（策略文件）](#)”和第 604 页的“[连接到套接字](#)”。

二进制套接字连接与 XML 套接字类似，不同之处是客户端与服务器不需要专门交换 XML 数据包。连接可以将数据作为二进制信息传输。这样，您就可以连接到各种各样的服务，包括邮件服务器（POP3、SMTP 和 IMAP）和新闻服务器（NNTP）。

Socket 类

ActionScript 3.0 中引入的 Socket 类使 ActionScript 可以建立套接字连接并读取和写入原始二进制数据。它与 XMLSocket 类相似，但没有指定接收和传输数据的格式。使用二进制协议的服务器互操作时，Socket 类非常有用。使用二进制套接字连接，可以编写允许用一些不同的 Internet 协议（例如 POP3、SMTP、IMAP 和 NNTP）进行交互的代码。反过来，这又会使 Flash Player 能够连接到邮件和新闻服务器。

Flash Player 可通过使用服务器的二进制协议直接与该服务器连接。某些服务器使用 big-endian 字节顺序，某些服务器则使用 little-endian 字节顺序。Internet 上的大多数服务器使用 big-endian 字节顺序，因为“网络字节顺序”为 big-endian。little-endian 字节顺序很常用，因为 Intel® x86 体系结构使用该字节顺序。您应使用与收发数据的服务器的字节顺序相匹配的 endian 字节顺序。默认情况下，IDataInput 和 IDataOutput 接口执行的所有操作和实现这些接口的类（ByteArray、Socket 和 URLStream）都以 big-endian 格式编码；即，最高有效字节位于前面。这样做是为了匹配 Java 和官方网络字节顺序。要更改是使用 big-endian 还是使用 little-endian 字节顺序，可以将 endian 属性设置为 Endian.BIG_ENDIAN 或 Endian.LITTLE_ENDIAN。

 Socket 类继承 IDataInput 和 IDataOutput 接口（位于 flash.utils 包中）实现的所有方法，应使用这些方法从 Socket 读取数据和向其中写入数据。

XMLSocket 类

ActionScript 提供了一个内置的 XMLSocket 类，使用它可以打开与服务器的持续连接。这种打开的连接消除了反应时间问题，它通常用于实时的应用程序，例如聊天应用程序或多人游戏。传统的基于 HTTP 的聊天解决方案频繁轮询服务器，并使用 HTTP 请求来下载新的消息。与此相对照，XMLSocket 聊天解决方案保持与服务器的开放连接，这一连接允许服务器即时发送传入的消息，而无需客户端发出请求。

要创建套接字连接，必须创建服务器端应用程序来等待套接字连接请求，然后向 SWF 文件发送响应。可以使用 Java、Python 或 Perl 程序语言来编写这种类型的服务器端应用程序。若要使用 XMLSocket 类，服务器计算机必须运行可识别 XMLSocket 类使用的协议的守护程序。下面的列表说明了该协议：

- XML 消息通过全双工 TCP/IP 流套接字连接发送。
- 每个 XML 消息都是一个完整的 XML 文档，以一个零 (0) 字节结束。
- 通过一个 XMLSocket 连接发送和接收的 XML 消息的数量没有限制。

XMLSocket 类不能自动穿过防火墙，因为 XMLSocket 没有 HTTP 隧道功能（这与实时消息传递协议 (RTMP) 不同）。如果您需要使用 HTTP 隧道，应考虑改用 Flash Remoting 或 Flash Media Server（支持 RTMP）。

注：将服务器设置为与 XMLSocket 对象进行通讯可能会遇到一些困难。如果您的应用程序不需要进行实时交互，请使用 URLLoader 类，而不要使用 XMLSocket 类。

可以使用 XMLSocket 类的 XMLSocket.connect() 和 XMLSocket.send() 方法，通过套接字连接与服务器之间传输 XML。XMLSocket.connect() 方法与 Web 服务器端口建立套接字连接。XMLSocket.send() 方法将 XML 对象传递给套接字连接中指定的服务器。

调用 XMLSocket.connect() 方法时，Flash Player 打开与服务器的 TCP/IP 连接，并使该连接保持打开状态，直到发生以下任一事件：

- XMLSocket 类的 XMLSocket.close() 方法被调用。
- 对 XMLSocket 对象的引用不再存在。
- Flash Player 退出。
- 连接中断（例如，调制解调器断开连接）。

创建并连接到 Java XML 套接字服务器

下面的代码说明一个用 Java 编写的简单 XMLSocket 服务器，该服务器接受传入连接并在命令提示窗口中显示收到的消息。虽然从命令行启动服务器时可以指定其它端口号，但默认情况下，在本地计算机上的 8080 端口创建新服务器。

新建一个文本文档并添加下面的代码：

```
import java.io.*;
import java.net.*;

class SimpleServer
{
    private static SimpleServer server;
    ServerSocket socket;
    Socket incoming;
    BufferedReader readerIn;
    PrintStream printOut;

    public static void main(String[] args)
    {
        int port = 8080;

        try
        {
            port = Integer.parseInt(args[0]);
        }
        catch (ArrayIndexOutOfBoundsException e)
        {
            // Catch exception and keep going.
        }

        server = new SimpleServer(port);
    }

    private SimpleServer(int port)
    {
        System.out.println(">> Starting SimpleServer");
        try
        {
            socket = new ServerSocket(port);
            incoming = socket.accept();
            readerIn = new BufferedReader(new InputStreamReader(incoming.getInputStream()));
            printOut = new PrintStream(incoming.getOutputStream());
            printOut.println("Enter EXIT to exit.\r");
            out("Enter EXIT to exit.\r");
            boolean done = false;
            while (!done)
            {
                String str = readerIn.readLine();
                if (str == null)
                {
                    done = true;
                }
                else
                {
                    out("Echo: " + str + "\r");
                }
            }
        }
    }
}
```

```

        if(str.trim().equals("EXIT"))
        {
            done = true;
        }
    }
    incoming.close();
}
catch (Exception e)
{
    System.out.println(e);
}
}

private void out(String str)
{
    printOut.println(str);
    System.out.println(str);
}
}

```

将文档保存到硬盘，命名为 `SimpleServer.java` 并使用 Java 编译器对其进行编译，这会创建一个名为 `SimpleServer.class` 的 Java 类文件。

您可以通过打开命令提示并键入 `java SimpleServer` 来启动 XMLSocket 服务器。`SimpleServer.class` 文件可以位于本地计算机或网络上的任何位置，不需要放置在 Web 服务器的根目录中。

 如果由于文件没有位于 Java 类路径中而无法启动服务器，请尝试使用 `java -classpath .SimpleServer` 启动服务器。

要从 ActionScript 应用程序连接到 XMLSocket，需要新建一个 XMLSocket 类实例，并在传递主机名和端口号时调用 `XMLSocket.connect()` 方法，如下所示：

```
var xmlsock:XMLSocket = new XMLSocket();
xmlsock.connect("127.0.0.1", 8080);
```

只要从服务器接收数据，就会调度 `data` 事件 (`flash.events.DataEvent.DATA`)：

```
xmlsock.addEventListener(DataEvent.DATA, onData);
private function onData(event:DataEvent):void
{
    trace("[ " + event.type + " ] " + event.data);
}
```

若要将数据发送到 XMLSocket 服务器，可以使用 `XMLSocket.send()` 方法并传递 XML 对象或字符串。Flash Player 将提供的参数转换为 `String` 对象，并将内容发送到 XMLSocket 服务器（后跟零 (0) 字节）：

```
xmlsock.send(xmlFormattedData);
```

`XMLSocket.send()` 方法不返回指示数据是否成功传输的值。如果尝试发送数据时发生错误，将引发 `IOError` 错误。

 发送到 XML 套接字服务器的每条消息必须以换行符 (`\n`) 结束。

存储本地数据

共享对象（有时称为“Flash cookie”）是一个数据文件，您访问的站点可能会在您的计算机上创建该文件。共享对象通常用于增强您的 Web 浏览体验 — 例如，使用这种对象可以个性化经常访问的网站的外观。共享对象本身不能对您计算机上的数据进行任何操作，也不能使用这些数据进行任何操作。更重要的是，共享对象永远不能访问或记住您的电子邮件地址或其它个人信息 — 除非您愿意提供这样的信息。

可以使用静态的 SharedObject.getLocal() 或 SharedObject.getRemote() 方法来创建新的共享对象实例。getLocal() 方法尝试加载仅对当前客户端可用的本地永久共享对象，而 getRemote() 方法则尝试加载可借助服务器（例如 Flash Media Server）跨多个客户端共享的远程共享对象。如果本地或远程共享对象不存在，则 getLocal() 和 getRemote() 方法将创建一个新的 SharedObject 实例。

下面的代码尝试加载名为 test 的本地共享对象。如果此共享对象不存在，则会新建一个具有此名称的共享对象。

```
var so:SharedObject = SharedObject.getLocal("test");
trace("SharedObject is " + so.size + " bytes");
```

如果找不到名为 test 的共享对象，则新建一个 0 字节大小的共享对象。如果已经存在该共享对象，则返回其当前大小（以字节为单位）。

您可以通过向数据对象分配值来在共享对象中存储数据，如以下示例所示：

```
var so:SharedObject = SharedObject.getLocal("test");
so.data.now = new Date().time;
trace(so.data.now);
trace("SharedObject is " + so.size + " bytes");
```

如果已经存在具有名称 test 和参数 now 的共享对象，则会覆盖现有值。您可以使用 SharedObject.size 属性来确定共享对象是否已存在，如以下示例所示：

```
var so:SharedObject = SharedObject.getLocal("test");
if (so.size == 0)
{
    // Shared object doesn't exist.
    trace("created...");
    so.data.now = new Date().time;
}
trace(so.data.now);
trace("SharedObject is " + so.size + " bytes");
```

上面的代码使用 size 参数确定具有指定名称的共享对象实例是否已存在。如果测试下面的代码，您会注意到每次运行该代码时，都会重新创建共享对象。为了将共享对象保存到用户的硬盘驱动器，您必须显式地调用 SharedObject.flush() 方法，如下面的示例所示：

```
var so:SharedObject = SharedObject.getLocal("test");
if (so.size == 0)
{
    // Shared object doesn't exist.
    trace("created...");
    so.data.now = new Date().time;
}
trace(so.data.now);
trace("SharedObject is " + so.size + " bytes");
so.flush();
```

使用 flush() 方法将共享对象写入用户的硬盘驱动器时，应仔细检查用户是否已使用 Flash Player 设置管理器（www.macromedia.com/support/documentation/en/flashplayer/help/settings_manager07.html）显式禁用了本地存储，如下面的示例所示：

```
var so:SharedObject = SharedObject.getLocal("test");
trace("Current SharedObject size is " + so.size + " bytes.");
so.flush();
```

可以在共享对象的 data 属性中指定属性名称以从共享对象检索值。例如，如果运行下面的代码，Flash Player 将显示 SharedObject 实例是在多少分钟前创建的：

```

var so:SharedObject = SharedObject.getLocal("test");
if (so.size == 0)
{
    // Shared object doesn't exist.
    trace("created...");
    so.data.now = new Date().time;
}
var ageMS:Number = new Date().time - so.data.now;
trace("SharedObject was created " + Number(ageMS / 1000 / 60).toPrecision(2) + " minutes ago");
trace("SharedObject is " + so.size + " bytes");
so.flush();

```

第一次运行上面的代码时，会创建一个名为 test 的新 SharedObject 实例，其初始大小为 0 字节。由于初始大小为 0 字节， if 语句的计算结果为 true，且一个名为 now 的新属性会添加到共享对象中。可以用当前时间减去 now 属性的值来计算共享对象的存在时间。以后每次运行上面的代码时，共享对象的大小都应大于 0，且该代码将跟踪共享对象是在多少分钟前创建的。

显示共享对象的内容

值存储在共享对象中的 data 属性中。可以使用 for..in 循环来循环访问共享对象中的每个值，如下面的示例所示：

```

var so:SharedObject = SharedObject.getLocal("test");
so.data.hello = "world";
so.data.foo = "bar";
so.data.timezone = new Date().timezoneOffset;
for (var i:String in so.data)
{
    trace(i + ":\t" + so.data[i]);
}

```

创建安全 SharedObject

当使用 getLocal() 或 getRemote() 创建本地或远程 SharedObject 时，有一个名为 secure 的可选参数，该参数确定对此共享对象的访问是否限于通过 HTTPS 连接传递的 SWF 文件。如果此参数设置为 true 且 SWF 文件是通过 HTTPS 传递的，Flash Player 将新建一个安全共享对象，或者获取对现有安全共享对象的引用。只可由通过 HTTPS 传递的 SWF 文件对此安全共享对象进行读取或写入，而该 HTTPS 调用 SharedObject.getLocal() 并将 secure 参数设置为 true。如果此参数设置为 false 且 SWF 文件是通过 HTTPS 传递的，Flash Player 将新建一个共享对象，或者获取对现有共享对象的引用。

仅可由通过非 HTTPS 连接传递的 SWF 文件对此共享对象进行读取或写入。如果 SWF 文件是通过非 HTTPS 连接传递的，并且您尝试将此参数设置为 true，将无法创建新的共享对象（或访问以前创建的安全共享对象），并会引发错误，并且共享对象设置为 null。如果尝试通过非 HTTPS 连接运行以下代码片段，SharedObject.getLocal() 方法将引发错误：

```

try
{
    var so:SharedObject = SharedObject.getLocal("contactManager", null, true);
}
catch (error:Error)
{
    trace("Unable to create SharedObject.");
}

```

无论此参数为何值，创建的共享对象的数量都接近域所允许的磁盘空间的总量。

使用数据文件

FileReference 对象表示客户端或服务器计算机上的数据文件。通过 FileReference 类的方法可使应用程序本地加载和保存数据文件，以及与远程服务器之间传输文件数据。

FileReference 类为加载、传输和保存数据文件提供了两种不同的方法。自从引入 **FileReference** 类以来，该类已加入了 **browse()** 方法（让用户选择文件）、**upload()** 方法（将文件数据传输到远程服务器）和 **download()** 方法（从服务器检索这些数据并将其保存在本地文件中）。从 Flash Player 10 和 Adobe AIR 1.5 开始，**FileReference** 类拥有了 **load()** 和 **save()** 方法，使您也可以直接访问和保存本地文件。这些方法的用法类似于 **URLLoader** 和 **Loader** 类中的同名方法。本节对 **FileReference** 类的这两种用法均加以讨论。

注：AIR 运行时还提供了其它类（在 **flash.filesystem** 包中）用于处理文件和本地文件系统。**flash.filesystem** 类提供的功能比 **FileReference** 类更多，但只有 AIR 才支持这些功能，而 Flash Player 则不支持。

FileReference 类

每个 **FileReference** 对象都引用本地计算机上的一个数据文件。**FileReference** 类的属性包含有关文件大小、类型、名称、文件扩展名、创建者、创建日期和修改日期的信息。

注：仅 Mac OS 支持 **creator** 属性。所有其它平台都会返回 **null**。

注：只有在 AIR 运行时中才支持 **extension** 属性。

可以通过以下两种方式之一创建 **FileReference** 类的实例：

- 使用 **new** 运算符，如下面的代码所示：

```
import flash.net.FileReference;
var fileRef:FileReference = new FileReference();
```

- 调用 **FileReferenceList.browse()** 方法，该方法将打开一个对话框，提示用户选择一个或多个要上载的文件。如果用户成功选择了一个或多个文件，则创建一个 **FileReference** 对象数组。

在创建完 **FileReference** 对象后，您便可以进行以下操作：

- 调用 **FileReference.browse()** 方法，该方法将打开一个对话框，提示用户从本地文件系统选择一个文件。发生这种情况之后，接下来通常要调用 **FileReference.upload()** 方法，将文件上传到远程服务器，或调用 **FileReference.load()** 方法，打开本地文件。
- 调用 **FileReference.download()** 方法。这会打开一个对话框，让用户选择用于保存新文件的位置。然后从服务器下载数据，并将数据存储在新文件中。
- 调用 **FileReference.load()** 方法。此方法使用 **browse()** 方法开始从之前所选的文件中加载数据。直到 **browse()** 操作完成（用户选择了文件）时，才能调用 **load()** 方法。
- 调用 **FileReference.save()** 方法。此方法将打开一个对话框，提示用户在本地文件系统上选择一个文件位置。然后此方法将数据保存到该指定位置。

注：一次只能执行一个 **browse()**、**download()** 或 **save()** 操作，因为在任何时刻都只能打开一个对话框。

除非发生以下情况之一，否则不会填充 **FileReference** 对象属性（如 **name**、**size** 或 **modificationDate**）：

- 已调用 **FileReference.browse()** 方法或 **FileReferenceList.browse()** 方法，并且用户已经使用对话框选择了文件。
- 已调用 **FileReference.download()** 方法，并且用户已经使用对话框指定了新的文件位置。

注：当执行下载时，在下载完成前只填充 **FileReference.name** 属性。下载完文件后，所有属性都将可用。

在执行对 **FileReference.browse()**、**FileReferenceList.browse()**、**FileReference.download()**、**FileReference.load()** 或 **FileReference.save()** 方法的调用时，大多数播放器将继续播放 SWF 文件，其中包括调度事件和执行代码。

对于上载和下载操作，SWF 文件只能访问自己的域（包括由策略文件指定的任何域）内的文件。如果包含文件的服务器与启动上载或下载的 SWF 文件不在同一个域中，则需要在该服务器上放置策略文件。

从文件加载数据

使用 `FileReference.load()` 方法可以将数据从本地文件加载到内存中。代码必须先调用 `FileReference.browse()` 方法，让用户选择要加载的文件。

调用 `FileReference.load()` 方法后，该方法立即就会返回，但并不能立即得到所加载的数据。`FileReference` 对象在加载过程的每个步骤中都会调度事件以调用侦听器方法。

`FileReference` 对象在加载过程中将调度以下事件。

- `open` 事件 (`Event.OPEN`): 在加载操作开始时调度。
- `progress` 事件 (`ProgressEvent.PROGRESS`): 以字节为单位从文件读取数据时定期调度。
- `complete` 事件 (`Event.COMPLETE`): 加载操作成功完成时调度。
- `ioError` 事件 (`IOErrorEvent.IO_ERROR`): 如果由于在打开文件或从文件读取数据时发生输入 / 输出错误而使加载过程失败，则调度此事件。

`FileReference` 对象调度 `complete` 事件后，即可将所加载的数据作为 `FileReference` 对象的 `data` 属性中的 `ByteArray` 进行访问。

下面的示例演示如何提示用户选择一个文件，然后从该文件将数据加载到内存中：

```
package
{
    import flash.display.Sprite;
    import flash.events.*;
    import flash.net.FileFilter;
    import flash.net.FileReference;
    import flash.net.URLRequest;
    import flash.utils.ByteArray;

    public class FileReferenceExample1 extends Sprite
    {
        private var fileRef:FileReference;
        public function FileReferenceExample1()
        {
            fileRef = new FileReference();
            fileRef.addEventListener(Event.SELECT, onFileSelected);
            fileRef.addEventListener(Event.CANCEL, onCancel);
            fileRef.addEventListener(IOErrorEvent.IO_ERROR, onIOError);
            fileRef.addEventListener(SecurityErrorEvent.SECURITY_ERROR,
                onSecurityError);
            var fileTypeFilter:FileFilter = new FileFilter("Text Files (*.txt, *.rtf)",
                "*.*txt;*.rtf");
            fileRef.browse([fileTypeFilter]);
        }
        public function onFileSelected(evt:Event):void
        {
            fileRef.addEventListener(ProgressEvent.PROGRESS, onProgress);
            fileRef.addEventListener(Event.COMPLETE, onComplete);
            fileRef.load();
        }

        public function onProgress(evt:ProgressEvent):void
        {
            trace("Loaded " + evt.bytesLoaded + " of " + evt.bytesTotal + " bytes.");
        }

        public function onComplete(evt:Event):void
        
```

```

    {
        trace("File was successfully loaded.");
        trace(fileRef.data);
    }

    public function onCancel(evt:Event):void
    {
        trace("The browse request was canceled by the user.");
    }

    public function onIOError(evt:IOErrorEvent):void
    {
        trace("There was an IO Error.");
    }

    public function onSecurityError(evt:Event):void
    {
        trace("There was a security error.");
    }
}
}

```

示例代码首先创建名为 fileRef 的 FileReference 对象，然后调用其 browse() 方法。此时将打开一个对话框，提示用户选择文件。选择文件后，将调用 onFileSelected() 方法。此方法添加针对 progress 和 complete 事件的侦听器，然后调用 FileReference 对象的 load() 方法。示例中的其它处理函数方法只是输出消息以报告加载操作的进度。在加载完成时，应用程序将使用 trace() 方法显示所加载文件的内容。

将数据保存到本地文件

使用 FileReference.save() 方法可以将数据保存到本地文件。该方法在开始时会打开一个对话框，让用户输入新文件名以及用于保存文件的位置。在用户选择了文件名和位置后，数据会写入到新文件中。在成功保存文件之后，将使用本地文件的属性填充 FileReference 对象的属性。

注：只有在响应用户事件（如针对鼠标单击或按键事件的事件处理函数）时，您的代码才应调用 FileReference.save() 方法。否则将引发错误。

调用 FileReference.save() 方法之后，该方法立即就会返回。FileReference 对象随后在文件保存过程的每个步骤中调度事件以调用侦听器方法。

FileReference 对象在文件保存过程中可以调度以下事件：

- **select 事件 (Event.SELECT)**: 在用户为要保存的新文件指定位置和文件名时调度。
- **cancel 事件 (Event.CANCEL)**: 在用户单击对话框中的“取消”按钮时调度。
- **open 事件 (Event.OPEN)**: 在保存操作开始时调度。
- **progress 事件 (ProgressEvent.PROGRESS)**: 以字节为单位向文件保存数据时定期调度。
- **complete 事件 (Event.COMPLETE)**: 保存操作成功完成时调度。
- **ioError 事件 (IOErrorEvent.IO_ERROR)**: 如果由于尝试向文件保存数据时发生输入 / 输出错误而使保存过程失败，则调度此事件。

FileReference.save() 方法的 data 参数中传递的对象类型决定着向文件写入数据的方式：

- 如果数据为 String 值，则使用 UTF-8 编码将其保存为文本文件。
- 如果数据为 XML 对象，则以 XML 格式将其写入到文件，并保留所有格式设置。
- 如果数据为 ByteArray 对象，则将其内容不进行任何转换地直接写入到文件。
- 如果数据为其它某种对象，则 FileReference.save() 方法调用该对象的 `toString()` 方法，然后将所得的 String 值保存到 UTF-8 文本文件。如果无法调用该对象的 `toString()` 方法，则会引发错误。

如果 `data` 参数的值为 `null`, 则会引发错误。

下面的代码扩展了前面针对 `FileReference.load()` 方法的示例。从文件读取数据后, 此示例提示用户提供文件名, 然后将数据保存在新文件中:

```
package
{
    import flash.display.Sprite;
    import flash.events.*;
    import flash.net.FileFilter;
    import flash.net.FileReference;
    import flash.net.URLRequest;
    import flash.utils.ByteArray;

    public class FileReferenceExample2 extends Sprite
    {
        private var fileRef:FileReference;
        public function FileReferenceExample2()
        {
            fileRef = new FileReference();
            fileRef.addEventListener(Event.SELECT, onFileSelected);
            fileRef.addEventListener(Event.CANCEL, onCancel);
            fileRef.addEventListener(IOErrorEvent.IO_ERROR, onIOError);
            fileRef.addEventListener(SecurityErrorEvent.SECURITY_ERROR,
                onSecurityError);
            var textTypeFilter:FileFilter = new FileFilter("Text Files (*.txt, *.rtf)",
                "*.*txt;*.rtf");
            fileRef.browse([textTypeFilter]);
        }
        public function onFileSelected(evt:Event):void
        {
            fileRef.addEventListener(ProgressEvent.PROGRESS, onProgress);
            fileRef.addEventListener(Event.COMPLETE, onComplete);
            fileRef.load();
        }

        public function onProgress(evt:ProgressEvent):void
        {
            trace("Loaded " + evt.bytesLoaded + " of " + evt.bytesTotal + " bytes.");
        }
        public function onCancel(evt:Event):void
        {
            trace("The browse request was canceled by the user.");
        }
        public function onComplete(evt:Event):void
        {
            trace("File was successfully loaded.");
            fileRef.removeEventListener(Event.SELECT, onFileSelected);
            fileRef.removeEventListener(ProgressEvent.PROGRESS, onProgress);
            fileRef.removeEventListener(Event.COMPLETE, onComplete);
            fileRef.removeEventListener(Event.CANCEL, onCancel);
            saveFile();
        }
        public function saveFile():void
        {
            fileRef.addEventListener(Event.SELECT, onSaveFileSelected);
            fileRef.save(fileRef.data,"NewFileName.txt");
        }

        public function onSaveFileSelected(evt:Event):void
        {
            fileRef.addEventListener(ProgressEvent.PROGRESS, onSaveProgress);
        }
    }
}
```

```

        fileRef.addEventListener(Event.COMPLETE, onSaveComplete);
        fileRef.addEventListener(Event.CANCEL, onSaveCancel);
    }

    public function onSaveProgress(evt:ProgressEvent):void
    {
        trace("Saved " + evt.bytesLoaded + " of " + evt.bytesTotal + " bytes.");
    }

    public function onSaveComplete(evt:Event):void
    {
        trace("File saved.");
        fileRef.removeEventListener(Event.SELECT, onSaveFileSelected);
        fileRef.removeEventListener(ProgressEvent.PROGRESS, onSaveProgress);
        fileRef.removeEventListener(Event.COMPLETE, onSaveComplete);
        fileRef.removeEventListener(Event.CANCEL, onSaveCancel);
    }

    public function onSaveCancel(evt:Event):void
    {
        trace("The save request was canceled by the user.");
    }

    public function onIOError(evt:IOErrorEvent):void
    {
        trace("There was an IO Error.");
    }

    public function onSecurityError(evt:Event):void
    {
        trace("There was a security error.");
    }
}
}

```

从文件加载了所有数据后，将调用 `onComplete()` 方法。`onComplete()` 方法删除针对加载事件的侦听器，然后调用 `saveFile()` 方法。`saveFile()` 方法调用 `FileReference.save()` 方法，该方法会打开一个新的对话框，让用户输入新文件名和用于保存文件的位置。其余事件侦听器方法在文件保存过程完成之前跟踪该过程的进度。

将文件上载到服务器

要将文件上载到服务器，需要首先调用 `browse()` 方法，以允许用户选择一个或多个文件。接下来，当调用 `FileReference.upload()` 方法时，将所选文件传输到服务器。如果用户使用 `FileReferenceList.browse()` 方法选择了多个文件，则 Flash Player 会创建所选文件的数组，称为 `FileReferenceList.fileList`。随后便可使用 `FileReference.upload()` 方法分别上载每个文件。

注： 使用 `FileReference.browse()` 方法时，您只能上载单个文件。要允许用户上载多个文件，您必须使用 `FileReferenceList.browse()` 方法。

虽然开发人员可以通过使用 `FileFilter` 类并将文件过滤器实例数组传递给 `browse()` 方法来指定一个或多个自定义文件类型过滤器，但是默认情况下，系统文件选取器对话框允许用户从本地计算机选取任何文件类型：

```

var imageTypes:FileFilter = new FileFilter("Images (*.jpg, *.jpeg, *.gif, *.png)", "*.jpg; *.jpeg; *.gif;
*.png");
var textTypes:FileFilter = new FileFilter("Text Files (*.txt, *.rtf)", "*.txt; *.rtf");
var allTypes:Array = new Array(imageTypes, textTypes);
var fileRef:FileReference = new FileReference();
fileRef.browse(allTypes);

```

用户在系统文件选取器中选择文件并单击“打开”按钮后，会调度 `Event.SELECT` 事件。如果使用 `FileReference.browse()` 方法选择要上载的文件，则需要使用以下代码将该文件发送到 Web 服务器：

```

var fileRef:FileReference = new FileReference();
fileRef.addEventListener(Event.SELECT, selectHandler);
fileRef.addEventListener(Event.COMPLETE, completeHandler);
try
{
    var success:Boolean = fileRef.browse();
}
catch (error:Error)
{
    trace("Unable to browse for files.");
}
function selectHandler(event:Event):void
{
    var request:URLRequest = new URLRequest("http://www.[yourdomain].com/fileUploadScript.cfm")
    try
    {
        fileRef.upload(request);
    }
    catch (error:Error)
    {
        trace("Unable to upload file.");
    }
}
function completeHandler(event:Event):void
{
    trace("uploaded");
}

```

 通过使用 URLRequest.method 和 URLRequest.data 属性以 POST 或 GET 方法发送变量，可以用 FileReference.upload() 方法将数据发送到服务器。

尝试使用 FileReference.upload() 方法上载文件时，可能会调度以下事件：

- open 事件 (Event.OPEN): 在上载操作开始时调度。
- progress 事件 (ProgressEvent.PROGRESS): 以字节为单位上载文件中的数据时定期调度。
- complete 事件 (Event.COMPLETE): 上载操作成功完成时调度。
- httpStatus 事件 (HTTPStatusEvent.HTTP_STATUS): 上载过程因 HTTP 错误而失败时调度。
- httpResponseStatus 事件 (HTTPStatusEvent.HTTP_RESPONSE_STATUS): 如果调用 upload() 或 uploadUnencoded() 方法时尝试通过 HTTP 访问数据，并且 Adobe AIR 可以检测并返回请求的状态代码，则调度此事件。
- securityError 事件 (SecurityErrorEvent.SECURITY_ERROR): 上载操作因违反安全规则而失败时调度。
- uploadCompleteData 事件 (DataEvent.UPLOAD_COMPLETE_DATA): 成功上载并从服务器接收数据之后调度。
- ioError 事件 (IOErrorEvent.IO_ERROR): 由于下列任何原因导致上载过程失败时调度：
 - 当 Flash Player 正在读取、写入或传输文件时发生输入 / 输出错误。
 - SWF 尝试将文件上载到要求身份验证（如用户名和密码）的服务器。在上载期间，Flash Player 不提供用户用于输入密码的方法。
 - url 参数包含无效协议。FileReference.upload() 方法必须使用 HTTP 或 HTTPS。

 Flash Player 不对需要身份验证的服务器提供完全支持。只有使用浏览器插件或 Microsoft ActiveX® 控件在浏览器中运行的 SWF 文件才可以提供一个对话框，来提示用户输入用户名和密码以进行身份验证，并且仅用于下载操作。对于使用插件或 ActiveX 控件进行的上载操作，或者使用独立或外部播放器进行的上载 / 下载操作，文件传输会失败。

若要以 ColdFusion 创建服务器脚本以接受来自 Flash Player 的文件上载，可以使用类似于以下内容的代码：

```
<cffile action="upload" filefield="Filedata" destination="#ExpandPath('./')#" nameconflict="OVERWRITE" />
```

此 ColdFusion 代码上载 Flash Player 发送的文件，并将其保存到 ColdFusion 模板所在的目录中以覆盖具有相同名称的任何文件。上面的代码显示接受文件上载所需的最低代码量；不应在生产环境中使用此脚本。理想情况下，应添加数据验证，以确保用户只上载接受的文件类型（例如图像），而不是上载可能危险的服务器端脚本。

下面的代码使用 PHP 说明文件上载，并且包含数据验证。该脚本将上载目录中的上载文件数限制为 10，确保文件小于 200 KB，并且只允许 JPEG、GIF 或 PNG 文件上载和保存到文件系统。

```
<?php
$MAXIMUM_FILESIZE = 1024 * 200; // 200KB
$MAXIMUM_FILE_COUNT = 10; // keep maximum 10 files on server
echo exif_imagetype($_FILES['Filedata']);
if ($_FILES['Filedata']['size'] <= $MAXIMUM_FILESIZE)
{
    move_uploaded_file($_FILES['Filedata']['tmp_name'], "./temporary/".$_FILES['Filedata']['name']);
    $type = exif_imagetype("./temporary/".$_FILES['Filedata']['name']);
    if ($type == 1 || $type == 2 || $type == 3)
    {
        rename("./temporary/".$_FILES['Filedata']['name'], "./images/".$_FILES['Filedata']['name']);
    }
    else
    {
        unlink("./temporary/".$_FILES['Filedata']['name']);
    }
}
$directory = opendir('./images/');
$files = array();
while ($file = readdir($directory))
{
    array_push($files, array('./images/'.$file, filectime('./images/'.$file)));
}
usort($files, sorter);
if (count($files) > $MAXIMUM_FILE_COUNT)
{
    $files_to_delete = array_splice($files, 0, count($files) - $MAXIMUM_FILE_COUNT);
    for ($i = 0; $i < count($files_to_delete); $i++)
    {
        unlink($files_to_delete[$i][0]);
    }
}
print_r($files);
closedir($directory);

function sorter($a, $b)
{
    if ($a[1] == $b[1])
    {
        return 0;
    }
    else
    {
        return ($a[1] < $b[1]) ? -1 : 1;
    }
}
?>
```

可以使用 POST 或 GET 请求方法将附加变量传递到上载脚本。要将附加 POST 变量发送到上载脚本，可以使用下面的代码：

```
var fileRef:FileReference = new FileReference();
fileRef.addEventListener(Event.SELECT, selectHandler);
fileRef.addEventListener(Event.COMPLETE, completeHandler);
fileRef.browse();
function selectHandler(event:Event):void
{
    var params:URLVariables = new URLVariables();
    params.date = new Date();
    params.ssid = "94103-1394-2345";
    var request:URLRequest = new
URLRequest("http://www.yourdomain.com/FileReferenceUpload/fileupload.cfm");
    request.method = URLRequestMethod.POST;
    request.data = params;
    fileRef.upload(request, "Custom1");
}
function completeHandler(event:Event):void
{
    trace("uploaded");
}
```

上面的示例新建一个将传递到远程服务器端脚本的 `URLVariables` 对象。在早期版本的 ActionScript 中，可以通过在查询字符串中传递值来将变量传递到服务器上载脚本。在 ActionScript 3.0 中，可以使用 `URLRequest` 对象将变量传递到远程脚本，该对象允许您使用 POST 或 GET 方法传递数据；因此，可以更轻松和更清晰地传递较大数据集。为了指定是使用 GET 还是使用 POST 请求方法来传递变量，可以将 `URLRequest.method` 属性相应设置为 `URLRequestMethod.GET` 或 `URLRequestMethod.POST`。

在 ActionScript 3.0 中，还可以通过向 `upload()` 方法提供第二个参数来覆盖默认 `Filedata` 上载文件字段名称，如上面的示例所示（该示例使用 `Custom1` 替换默认值 `Filedata`）。

默认情况下，Flash Player 不尝试发送测试上载，虽然您可以通过将值 `true` 作为第三个参数传递给 `upload()` 方法来覆盖此行为。测试上载的目的是检查实际文件上载是否会成功，如果需要服务器身份，还会检查服务器身份验证是否会成功。

注：目前，只在基于 Windows 的 Flash Player 上进行测试上载。

处理文件上载的服务器脚本应收到包含下列元素的 HTTP POST 请求：

- Content-Type，其值为 `multipart/form-data`。
- Content-Disposition，其 name 属性设置为“`Filedata`”，filename 属性设置为原始文件的名称。您可以通过在 `FileReference.upload()` 方法中传递 `uploadDataFieldName` 参数的值来指定自定义 name 属性。
- 文件的二进制内容。

下面是一个 HTTP POST 请求范例：

```
POST /handler.asp HTTP/1.1
Accept: text/*
Content-Type: multipart/form-data;
boundary=-----Ij5ae0ae0KM7GI3KM7ei4cH2ei4gL6
User-Agent: Shockwave Flash
Host: www.mydomain.com
Content-Length: 421
Connection: Keep-Alive
Cache-Control: no-cache

-----Ij5ae0ae0KM7GI3KM7ei4cH2ei4gL6
Content-Disposition: form-data; name="Filename"

sushi.jpg
-----Ij5ae0ae0KM7GI3KM7ei4cH2ei4gL6
Content-Disposition: form-data; name="Filedata"; filename="sushi.jpg"
Content-Type: application/octet-stream

Test File
-----Ij5ae0ae0KM7GI3KM7ei4cH2ei4gL6
Content-Disposition: form-data; name="Upload"

Submit Query
-----Ij5ae0ae0KM7GI3KM7ei4cH2ei4gL6
(actual file data,,)
```

下面的 HTTP POST 请求范例发送三个 POST 变量: `api_sig`、`api_key` 和 `auth_token`, 并使用自定义上载数据字段名的值 "photo":

```
POST /handler.asp HTTP/1.1
Accept: text/*
Content-Type: multipart/form-data;
boundary=-----Ij5ae0ae0KM7GI3KM7ei4cH2ei4gL6
User-Agent: Shockwave Flash
Host: www.mydomain.com
Content-Length: 421
Connection: Keep-Alive
Cache-Control: no-cache

-----Ij5GI3GI3ei4GI3ei4KM7GI3KM7KM7
Content-Disposition: form-data; name="Filename"

sushi.jpg
-----Ij5GI3GI3ei4GI3ei4KM7GI3KM7KM7
Content-Disposition: form-data; name="api_sig"

XXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
-----Ij5GI3GI3ei4GI3ei4KM7GI3KM7KM7
Content-Disposition: form-data; name="api_key"

XXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
-----Ij5GI3GI3ei4GI3ei4KM7GI3KM7KM7
Content-Disposition: form-data; name="auth_token"

XXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
-----Ij5GI3GI3ei4GI3ei4KM7GI3KM7KM7
Content-Disposition: form-data; name="photo"; filename="sushi.jpg"
Content-Type: application/octet-stream

(actual file data,,,
-----Ij5GI3GI3ei4GI3ei4KM7GI3KM7KM7
Content-Disposition: form-data; name="Upload"

Submit Query
-----Ij5GI3GI3ei4GI3ei4KM7GI3KM7KM7--
```

从服务器下载文件

您可以让用户使用 `FileReference.download()` 方法从服务器下载文件，该方法使用两个参数：`request` 和 `defaultFileName`。第一个参数是 `URLRequest` 对象，该对象包含要下载的文件的 URL。第二个参数是可选的，使用该参数可以指定出现在下载文件对话框中的默认文件名。如果省略第二个参数 `defaultFileName`，则使用指定 URL 中的文件名。

下面的代码从 SWF 文档所在的目录下载名为 `index.xml` 的文件：

```
var request:URLRequest = new URLRequest("index.xml");
var fileRef:FileReference = new FileReference();
fileRef.download(request);
```

若要将默认名称设置为 `currentnews.xml` 而非 `index.xml`，请指定 `defaultFileName` 参数，如下面的片断所示：

```
var request:URLRequest = new URLRequest("index.xml");
var fileToDownload:FileReference = new FileReference();
fileToDownload.download(request, "currentnews.xml");
```

如果服务器文件名不直观，或者文件名是由服务器生成的，则文件重命名可能非常有用。另外，在使用服务器端脚本下载文件（而不是直接下载文件）时，最好显式指定 `defaultFileName` 参数。例如，如果有基于传递给它的 URL 变量下载特定文件的服务器端脚本，则需要指定 `defaultFileName` 参数。否则，下载文件的默认名称即是服务器端脚本的名称。

可以使用 `download()` 方法将数据发送至服务器，方法是将参数追加到要分析的服务器脚本的 URL。下面的 ActionScript 3.0 片断基于传递给 ColdFusion 脚本的参数下载文档：

```

package
{
    import flash.display.Sprite;
    import flash.net.FileReference;
    import flash.net.URLRequest;
    import flash.net.URLRequestMethod;
    import flash.net.URLVariables;

    public class DownloadFileExample extends Sprite
    {
        private var fileToDownload:FileReference;
        public function DownloadFileExample()
        {
            var request:URLRequest = new URLRequest();
            request.url = "http://www.[yourdomain].com/downloadfile.cfm";
            request.method = URLRequestMethod.GET;
            request.data = new URLVariables("id=2");
            fileToDownload = new FileReference();
            try
            {
                fileToDownload.download(request, "file2.txt");
            }
            catch (error:Error)
            {
                trace("Unable to download file.");
            }
        }
    }
}

```

下面的代码显示了 ColdFusion 脚本 download.cfm，该脚本根据 URL 变量的值从服务器下载两个文件之一：

```

<cfparam name="URL.id" default="1" />
<cfswitch expression="#URL.id#">
    <cfcase value="2">
        <cfcontent type="text/plain" file="#ExpandPath('two.txt')#" deletefile="No" />
    </cfcase>
    <cfdefaultcase>
        <cfcontent type="text/plain" file="#ExpandPath('one.txt')#" deletefile="No" />
    </cfdefaultcase>
</cfswitch>

```

FileNotFoundException 类

使用 `FileNotFoundException` 类，用户可以选择一个或多个要上传到服务器端脚本的文件。文件上传是由 `FileReference.upload()` 方法处理的，必须对用户选择的每个文件调用此方法。

下面的代码创建两个 `FileFilter` 对象（`imageFilter` 和 `textFilter`），并在一个数组中将它们传递到 `FileNotFoundException.browse()` 方法。这导致操作系统文件对话框显示两个可能的文件类型过滤器。

```

var imageFilter:FileFilter = new FileFilter("Image Files (*.jpg, *.jpeg, *.gif, *.png)", "*.jpg; *.jpeg; *.gif; *.png");
var textFilter:FileFilter = new FileFilter("Text Files (*.txt, *.rtf)", "*.txt; *.rtf");
var fileRefList:FileNotFoundException = new FileNotFoundException();
try
{
    var success:Boolean = fileRefList.browse(new Array(imageFilter, textFilter));
}
catch (error:Error)
{
    trace("Unable to browse for files.");
}

```

允许用户使用 `FileReferenceList` 类选择并上载一个或多个文件与使用 `FileReference.browse()` 选择文件是相同的，但 `FileReferenceList` 允许选择多个文件。上载多个文件时，要求您使用 `FileReference.upload()` 上载每个所选的文件，如以下代码所示：

```
var fileRefList:FileReferenceList = new FileReferenceList();
fileRefList.addEventListener(Event.SELECT, selectHandler);
fileRefList.browse();

function selectHandler(event:Event):void
{
    var request:URLRequest = new URLRequest("http://www.[yourdomain].com/fileUploadScript.cfm");
    var file:FileReference;
    var files:FileReferenceList = FileReferenceList(event.target);
    var selectedFileArray:Array = files.fileList;
    for (var i:uint = 0; i < selectedFileArray.length; i++)
    {
        file = FileReference(selectedFileArray[i]);
        file.addEventListener(Event.COMPLETE, completeHandler);
        try
        {
            file.upload(request);
        }
        catch (error:Error)
        {
            trace("Unable to upload files.");
        }
    }
}
function completeHandler(event:Event):void
{
    trace("uploaded");
}
```

由于 `Event.COMPLETE` 事件会添加到数组中每个单独的 `FileReference` 对象中，因此，Flash Player 在每个文件完成上载时都会调用 `completeHandler()` 方法。

示例：构建 Telnet 客户端

Telnet 示例说明了使用 `Socket` 类连接远程服务器和传输数据的方法。该示例演示下列方法：

- 使用 `Socket` 类创建自定义 telnet 客户端
- 使用 `ByteArray` 对象将文本发送到远程服务器
- 处理从远程服务器收到的数据

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可以在 Samples/Telnet 文件夹中找到 Telnet 应用程序文件。该应用程序包含以下文件：

文件	描述
TelnetSocket.fla 或 TelnetSocket.mxml	由用户界面组成的 Flex 或 Flash 的主应用程序文件（分别为 MXML 和 FLA 格式）。
TelnetSocket.as	用于提供用户界面逻辑的文档类（仅限 Flash）。
com/example/programmingas3/Telnet/Telnet.as	为应用程序提供 Telnet 客户端功能，例如连接到远程服务器以及发送、接收和显示数据。

Telnet 套接字应用程序概述

主 TelnetSocket.mxml 文件负责为整个应用程序创建用户界面 (UI)。

除了 UI，此文件还定义两个方法 login() 和 sendCommand()，以便将用户连接到指定的服务器。

下面的代码列出了主应用程序文件中的 ActionScript：

```
import com.example.programmingas3.socket.Telnet;

private var telnetClient:Telnet;
private function connect():void
{
    telnetClient = new Telnet(serverName.text, int(portNumber.text), output);
    console.title = "Connecting to " + serverName.text + ":" + portNumber.text;
    console.enabled = true;
}
private function sendCommand():void
{
    var ba:ByteArray = new ByteArray();
    ba.writeMultiByte(command.text + "\n", "UTF-8");
    telnetClient.writeBytesToSocket(ba);
    command.text = "";
}
```

第一行代码从自定义 com.example.programmingas.socket 包中导入 Telnet 类。第二行代码声明 Telnet 类的实例 telnetClient，稍后 connect() 方法将初始化该实例。接下来，声明 connect() 方法，该方法初始化先前声明的 telnetClient 变量。此方法传递用户指定的 telnet 服务器名称、telnet 服务器端口和对显示列表中 TextArea 组件的引用，其中，显示列表用于显示来自套接字服务器的文本响应。connect() 方法的最后两行设置 Panel 的 title 属性并启用 Panel 组件，该组件允许用户将数据发送到远程服务器。主应用程序文件中的最后一个方法 sendCommand() 用于将用户的命令作为 ByteArray 对象发送到远程服务器。

Telnet 类概述

Telnet 类负责连接到远程 Telnet 服务器和发送 / 接收数据。

Telnet 类声明下列私有变量：

```
private var serverURL:String;
private var portNumber:int;
private var socket:Socket;
private var ta:TextArea;
private var state:int = 0;
```

第一个变量 serverURL 包含要连接到的用户指定的服务器地址。

第二个变量 portNumber 是 Telnet 服务器当前在其上运行的端口号。默认情况下，Telnet 服务在端口 23 上运行。

第三个变量 socket 是 Socket 实例，它将尝试连接到由 serverURL 和 portNumber 变量所定义的服务器。

第四个变量 `ta` 是对舞台上的 `TextArea` 组件实例的引用。此组件用于显示来自远程 Telnet 服务器的响应或者任何可能的错误消息。

最后一个变量 `state` 是数值，用于确定 Telnet 客户端支持哪些选项。

正如您之前所见，`Telnet` 类的构造函数由主应用程序文件中的 `connect()` 方法调用。

`Telnet` 构造函数采用三个参数：`server`、`port` 和 `output`。`server` 和 `port` 参数指定 Telnet 服务器在其上运行的服务器名称和端口号。最后一个参数 `output` 是对舞台上的 `TextArea` 组件实例的引用，将在舞台上向用户显示服务器输出。

```
public function Telnet(server:String, port:int, output:TextArea)
{
    serverURL = server;
    portNumber = port;
    ta = output;
    socket = new Socket();
    socket.addEventListener(Event.CONNECT, connectHandler);
    socket.addEventListener(Event.CLOSE, closeHandler);
    socket.addEventListener(ErrorEvent.ERROR, errorHandler);
    socket.addEventListener(IOErrorEvent.IO_ERROR, ioErrorHandler);
    socket.addEventListener(ProgressEvent.SOCKET_DATA, dataHandler);
    Security.loadPolicyFile("http://" + serverURL + "/crossdomain.xml");
    try
    {
        msg("Trying to connect to " + serverURL + ":" + portNumber + "\n");
        socket.connect(serverURL, portNumber);
    }
    catch (error:Error)
    {
        msg(error.message + "\n");
        socket.close();
    }
}
```

向套接字写入数据

要向套接字连接写入数据，可以调用 `Socket` 类中的任何写入方法（例如 `writeBoolean()`、`writeByte()`、`writeBytes()` 或 `writeDouble()`），然后使用 `flush()` 方法刷新输出缓冲区中的数据。在 Telnet 服务器中，使用 `writeBytes()` 方法向套接字连接中写入数据，该方法将字节数组作为参数，并将其发送到输出缓冲区。`writeBytesToSocket()` 方法如下：

```
public function writeBytesToSocket(ba:ByteArray):void
{
    socket.writeBytes(ba);
    socket.flush();
}
```

此方法是由主应用程序文件中的 `sendCommand()` 方法调用的。

显示来自套接字服务器的消息

当从套接字服务器接收消息，或者发生事件时，将调用自定义 `msg()` 方法。此方法将一个字符串追加到舞台上的 `TextArea`，并调用自定义 `setScroll()` 方法，该方法导致 `TextArea` 组件滚动到底端。`msg()` 方法如下：

```
private function msg(value:String):void
{
    ta.text += value;
    setScroll();
}
```

如果您未将 `TextArea` 组件的内容设置为自动滚动，则用户需要手动拖动文本区域中的滚动条才能看到来自服务器的最新响应。

滚动 TextArea 组件

`setScroll()` 方法包含一行 ActionScript，用于垂直滚动 TextArea 组件内容，以便用户可以查看返回文本的最后一行。下面的片断显示了 `setScroll()` 方法：

```
public function setScroll():void
{
    ta.verticalScrollPosition = ta.maxVerticalScrollPosition;
}
```

此方法设置 `verticalScrollPosition` 属性（该属性是当前显示的最上面一行字符的行号），并将其设置为 `maxVerticalScrollPosition` 属性的值。

示例：上载和下载文件

FileIO 示例说明了在 Flash Player 中执行文件下载和上载的方法。这些方法包括：

- 将文件下载到用户的计算机
- 将文件从用户的计算机上载到服务器
- 取消正在进行的下载
- 取消正在进行的上载

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。在 Samples/FileIO 文件夹中可以找到 FileIO 应用程序文件。该应用程序包含以下文件：

文件	描述
FileIO.fla 或 FileIO.mxml	Flash 或 Flex 中的主应用程序文件（分别为 FLA 和 MXML）。
com/example/programmingas3/fileio/FileDownload.as	一个类，包含用于从服务器下载文件的方法。
com/example/programmingas3/fileio/FileUpload.as	一个类，包含用于将文件上载到服务器的方法。

FileIO 应用程序概述

FileIO 应用程序包含用户界面，使用户可以使用 Flash Player 上载或下载文件。该应用程序首先定义一对自定义组件 `FileUpload` 和 `FileDownload`，可以在 `com.example.programmingas3.fileio` 包中找到这两个组件。每个自定义组件调度其 `contentComplete` 事件后，会调用该组件的 `init()` 方法，此方法传递对 `ProgressBar` 和 `Button` 组件实例的引用，从而使用户可以看到文件的上载或下载进度或者取消正在进行的文件传输。

FileIO.mxml 文件中的相关代码如下所示（请注意，在 Flash 版本中，FLA 文件包含放在舞台上的组件，这些组件的名称与此步骤中介绍的 Flex 组件的名称相匹配）：

```
<example:FileUpload id="fileUpload" creationComplete="fileUpload.init(uploadProgress, cancelUpload);" />
<example:FileDownload id="fileDownload" creationComplete="fileDownload.init(downloadProgress,
cancelDownload);" />
```

下面的代码显示“上载文件”面板，其中包含一个进度条和两个按钮。第一个按钮 `startUpload` 调用 `FileUpload.startUpload()` 方法，该方法调用 `FileReference.browse()` 方法。下面的摘选显示了用于“上载文件”面板的代码：

```
<mx:Panel title="Upload File" paddingTop="10" paddingBottom="10" paddingLeft="10" paddingRight="10">
    <mx:ProgressBar id="uploadProgress" label="" mode="manual" />
    <mx:ControlBar horizontalAlign="right">
        <mx:Button id="startUpload" label="Upload..." click="fileUpload.startUpload();"/>
        <mx:Button id="cancelUpload" label="Cancel" click="fileUpload.cancelUpload();" enabled="false" />
    </mx:ControlBar>
</mx:Panel>
```

此代码将一个 **ProgressBar** 组件实例和两个 **Button** 组件按钮实例放在舞台上。当用户单击“上载”按钮 (**startUpload**) 时，会启动一个操作系统对话框，允许用户选择要上传到远程服务器的文件。尽管用户开始文件上传时，另一个按钮 **cancelUpload** 会变为可用，并允许用户随时中断文件传输，但默认情况下禁用该按钮。

用于“下载文件”面板的代码如下：

```
<mx:Panel title="Download File" paddingTop="10" paddingBottom="10" paddingLeft="10" paddingRight="10">
    <mx:ProgressBar id="downloadProgress" label="" mode="manual" />
    <mx:ControlBar horizontalAlign="right">
        <mx:Button id="startDownload" label="Download..." click="fileDownload.startDownload();"/>
        <mx:Button id="cancelDownload" label="Cancel" click="fileDownload.cancelDownload();"
enabled="false" />
    </mx:ControlBar>
</mx:Panel>
```

此代码与文件上传代码非常相似。当用户单击“下载”按钮 (**startDownload**) 时，会调用 **FileDownload.startDownload()** 方法，从而开始下载 **FileDownload.DOWNLOAD_URL** 变量中指定的文件。在文件下载时，进度条进行更新，以显示文件下载百分比。用户可以随时通过单击 **cancelDownload** 按钮取消下载，这样会立即停止正在进行的文件下载。

从远程服务器下载文件

从远程服务器下载文件由 **flash.net.FileReference** 类和自定义 **com.example.programmingas3.fileio.FileDownload** 类处理。当用户单击“下载”按钮时，Flash Player 开始下载 **FileDownload** 类的 **DOWNLOAD_URL** 变量中指定的文件。

FileDownload 类从定义 **com.example.programmingas3.fileio** 包中的 4 个变量开始，如以下代码所示：

```
/**
 * Hard-code the URL of file to download to user's computer.
 */
private const DOWNLOAD_URL:String = "http://www.yourdomain.com/file_to_download.zip";

/**
 * Create a FileReference instance to handle the file download.
 */
private var fr:FileReference;

/**
 * Define reference to the download ProgressBar component.
 */
private var pb:ProgressBar;

/**
 * Define reference to the "Cancel" button which will immediately stop
 * the current download in progress.
 */
private var btn:Button;
```

第一个变量 **DOWNLOAD_URL** 包含文件的路径，当用户单击主应用程序文件中的“下载”按钮时，该文件会下载到用户的计算机上。

第二个变量 **fr** 是 **FileReference** 对象，该对象在 **FileDownload.init()** 方法中进行初始化，用于处理远程文件到用户计算机的下载。

最后两个变量 pb 和 btn 包含对舞台上的 ProgressBar 和 Button 组件实例的引用，它们由 FileDownload.init() 方法进行初始化。

初始化 FileDownload 组件

通过在 FileDownload 类中调用 init() 方法对 FileDownload 组件进行初始化。此方法使用两个参数 pb 和 btn，它们分别是 ProgressBar 和 Button 组件实例。

用于 init() 方法的代码如下所示：

```
/**
 * Set references to the components, and add listeners for the OPEN,
 * PROGRESS, and COMPLETE events.
 */
public function init(pb:ProgressBar, btn:Button):void
{
    // Set up the references to the progress bar and cancel button,
    // which are passed from the calling script.
    this.pb = pb;
    this.btn = btn;

    fr = new FileReference();
    fr.addEventListener(Event.OPEN, openHandler);
    fr.addEventListener(ProgressEvent.PROGRESS, progressHandler);
    fr.addEventListener(Event.COMPLETE, completeHandler);
}
```

开始文件下载

当用户单击舞台上的 Download Button 组件实例时，startDownload() 方法启动文件下载进程。下面的摘选显示了 startDownload() 方法：

```
/**
 * Begin downloading the file specified in the DOWNLOAD_URL constant.
 */
public function startDownload():void
{
    var request:URLRequest = new URLRequest();
    request.url = DOWNLOAD_URL;
    fr.download(request);
}
```

首先，startDownload() 方法创建一个新的 URLRequest 对象，并将目标 URL 设置为 DOWNLOAD_URL 变量指定的值。接下来，调用 FileReference.download() 方法，将新创建的 URLRequest 对象作为参数传递。这会导致操作系统在用户计算机上显示一个对话框，提示用户选择一个位置以保存所请求的文档。用户选择了位置后，将调度 open 事件 (Event.OPEN) 并调用 openHandler() 方法。

openHandler() 方法设置 ProgressBar 组件的 label 属性的文本格式，并启用使用户可以立即停止正在进行的下载的“取消”按钮。openHandler() 方法如下所示：

```
/**
 * When the OPEN event has dispatched, change the progress bar's label
 * and enable the "Cancel" button, which allows the user to abort the
 * download operation.
 */
private function openHandler(event:Event):void
{
    pb.label = "DOWNLOADING %3%";
    btn.enabled = true;
}
```

监视文件的下载进度

当从远程服务器向用户计算机下载文件时，会定期调度 progress 事件 (ProgressEvent.PROGRESS)。每当调度 progress 事件时，会调用 progressHandler() 方法并更新舞台上的 ProgressBar 组件实例。用于 progressHandler() 方法的代码如下所示：

```
/**
 * While the file is downloading, update the progress bar's status.
 */
private function progressHandler(event:ProgressEvent):void
{
    pb.setProgress(event.bytesLoaded, event.bytesTotal);
}
```

进度事件包含两个属性 (bytesLoaded 和 bytesTotal)，这两个属性用于更新舞台上的 ProgressBar 组件。这样，用户可以了解已完成下载的文件比例以及尚未下载的文件比例。用户可以通过单击进度条下的“取消”按钮随时终止文件传输。

如果文件下载成功，complete 事件 (Event.COMPLETE) 将调用 completeHandler() 方法，该方法通知用户文件已完成下载并禁用“取消”按钮。用于 completeHandler() 方法的代码如下所示：

```
/**
 * Once the download has completed, change the progress bar's label one
 * last time and disable the "Cancel" button since the download is
 * already completed.
 */
private function completeHandler(event:Event):void
{
    pb.label = "DOWNLOAD COMPLETE";
    btn.enabled = false;
}
```

取消文件下载

用户可以通过单击舞台上的“取消”按钮随时终止文件传输和停止下载任何更多字节。下面的摘选显示了用于取消下载的代码：

```
/**
 * Cancel the current file download.
 */
public function cancelDownload():void
{
    fr.cancel();
    pb.label = "DOWNLOAD CANCELLED";
    btn.enabled = false;
}
```

首先，该代码立即停止文件传输以禁止继续下载任何数据。接下来，进度条的 label 属性进行更新，通知用户下载已成功取消。最后，禁用“取消”按钮，从而阻止用户再次尝试下载文件之前再次单击该按钮。

将文件上载到远程服务器上

文件上载进程与文件下载进程非常相似。FileUpload 类声明相同的四个变量，如以下代码所示：

```
private const UPLOAD_URL:String = "http://www.yourdomain.com/your_upload_script.cfm";
private var fr:FileReference;
private var pb:ProgressBar;
private var btn:Button;
```

与 FileDownload.DOWNLOAD_URL 变量不同，UPLOAD_URL 变量包含要从用户计算机上载文件的服务器端脚本的 URL。其余三个变量的作用与 FileDownload 类中的对应变量相同。

初始化 FileUpload 组件

FileUpload 组件包含 init() 方法，此方法从主应用程序调用。此方法使用两个参数 pb 和 btn，它们是对舞台上的 ProgressBar 和 Button 组件实例的引用。接下来，init() 方法初始化先前在 FileUpload 类中定义的 FileReference 对象。最后，此方法将四个事件侦听器分配给 FileReference 对象。用于 init() 方法的代码如下所示：

```
public function init(pb:ProgressBar, btn:Button):void
{
    this.pb = pb;
    this.btn = btn;

    fr = new FileReference();
    fr.addEventListener(Event.SELECT, selectHandler);
    fr.addEventListener(Event.OPEN, openHandler);
    fr.addEventListener(ProgressEvent.PROGRESS, progressHandler);
    fr.addEventListener(Event.COMPLETE, completeHandler);
}
```

开始文件上载

当用户单击舞台上的“上载”按钮时，启动文件上载，此操作将调用 FileUpload.startUpload() 方法。此方法调用 FileReference 类的 browse() 方法，从而导致操作系统显示一个系统对话框，以提示用户选择要上载到远程服务器的文件。下面的摘选显示了用于 startUpload() 方法的代码：

```
public function startUpload():void
{
    fr.browse();
}
```

在用户选择要上载的文件后，会调度 select 事件 (Event.SELECT)，从而导致调用 selectHandler() 方法。selectHandler() 方法新建一个 URLRequest 对象，并将 URLRequest.url 属性设置为先前在代码中定义的 UPLOAD_URL 常量的值。最后，FileReference 对象将所选文件上载到指定的服务器端脚本。用于 selectHandler() 方法的代码如下所示：

```
private function selectHandler(event:Event):void
{
    var request:URLRequest = new URLRequest();
    request.url = UPLOAD_URL;
    fr.upload(request);
}
```

FileUpload 类中的其余代码与 FileDownload 类中定义的代码相同。如果用户想要在任何时间点终止上载，可以单击“取消”按钮，该按钮将在进度条上设置标签并立即停止文件传输。每当调度 progress 事件 (ProgressEvent.PROGRESS) 时，都会更新进度条。同样，当完成上载时，进度条进行更新以通知用户文件已上载成功。然后，禁用“取消”按钮，直到用户开始新的文件传输。

第 28 章：客户端系统环境

本章说明如何与用户的系统进行交互，具体介绍了如何确定支持哪些功能，以及如何通过用户安装的输入法编辑器 (IME)（如果有）生成多语言 SWF 文件。还将介绍应用程序域的典型用法。

客户端系统环境基础知识

客户端系统环境简介

在构建更高级的 ActionScript 应用程序时，您可能需要了解有关用户操作系统的详细信息并访问用户操作系统的功能。客户端系统环境是 `flash.system` 包中的类集合，可通过这些类来访问系统级功能，例如：

- 确定执行 SWF 时所在的应用程序和安全域
- 确定用户的 Flash® Player 或 Adobe® AIR™ 实例的功能（如屏幕大小（分辨率）），以及确定某项功能（如 mp3 音频）是否可用
- 使用 IME 建立多语言站点
- 与 Flash Player 的容器（可能是 HTML 页或容器应用程序）或 AIR 的容器进行交互。
- 将信息保存到用户的剪贴板中

`flash.system` 包还包括 `IMEConversionMode` 和 `SecurityPanel` 类。这两个类分别包含与 IME 和 Security 类一起使用的静态常数。

常见客户端系统环境任务

本章介绍了通过 ActionScript 使用客户端系统的以下常见任务：

- 确定应用程序使用的内存数量
- 将文本复制到用户的剪贴板中
- 确定用户计算机的功能，例如：
 - 屏幕分辨率、颜色、DPI 以及像素高宽比
 - 操作系统
 - 支持音频流、视频流以及 mp3 回放
 - 安装的 Flash Player 是否为调试版
- 处理应用程序域：
 - 定义应用程序域
 - 将 SWF 文件的代码划分到应用程序域中
- 在应用程序中处理 IME：
 - 确定是否安装了 IME
 - 确定并设置 IME 转换模式
 - 对文本字段禁用 IME
 - 检测何时发生 IME 转换

重要概念和术语

以下参考列表包含将会在本章中使用的重要术语：

- 操作系统：计算机上运行的主程序（其它所有应用程序均运行在其中），如 Microsoft Windows、Mac OS X 或 Linux[®]。
- 剪贴板：用于保存复制或剪切的文本或项目的操作系统容器，可从中将项目粘贴到应用程序中。
- 应用程序域：用于将不同 SWF 文件中使用的类分开的机制，以便在 SWF 文件包含具有相同名称的不同类时，这些类不会彼此覆盖。
- IME (input method editor, 输入法编辑器)：用于通过标准键盘输入复杂字符或符号的程序（或操作系统工具）。
- 客户端系统：在编程术语中，“客户端”是指在单独的计算机上运行并由单个用户使用的应用程序部分（或整个应用程序）。“客户端系统”是指用户计算机上的基础操作系统。

完成本章中的示例

在学习本章时，您可能想要自己动手测试一些示例代码清单。本章中的所有代码清单均包括相应的 trace() 函数调用，用于写出所测试的值。要测试本章中的代码清单，请执行以下操作：

- 1 创建一个空的 Flash 文档。
- 2 在时间轴上选择一个关键帧。
- 3 打开“动作”面板，将代码清单复制到“脚本”窗格中。
- 4 通过“控制”>“测试影片”运行程序。

您将在“输出”面板中看到该代码清单的 trace() 函数的结果。

后面的某些代码清单更为复杂一些，并且是以类的形式编写的。要测试这些示例，请执行以下操作：

- 1 创建一个空的 Flash 文档并将它保存到您的计算机上。
- 2 创建一个新的 ActionScript 文件，并将它保存到 Flash 文档所在的目录中。文件名应与代码清单中的类的名称一致。例如，如果代码清单定义一个名为 SystemTest 的类，则使用名称 SystemTest.as 来保存 ActionScript 文件。
- 3 将代码清单复制到 ActionScript 文件中并保存该文件。
- 4 在 Flash 文档中，单击舞台或工作区的空白部分，以激活文档的“属性”检查器。
- 5 在“属性”检查器的“文档类”字段中，输入您从文本中复制的 ActionScript 类的名称。
- 6 使用“控制”>“测试影片”运行程序

您将在“输出”面板中看到该示例的结果。

用于测试示例代码清单的方法在第 29 页的“[测试本章内的示例代码清单](#)”中有详细的介绍。

使用 System 类

可以通过 System 类包含的一些方法和属性与用户的操作系统进行交互，并检索 Flash Player 或 AIR 的当前内存使用数据。还可以使用 System 类的方法和属性来侦听 imeComposition 事件、指示 Flash Player 或 AIR 使用用户的当前代码页加载外部文本文件或按 Unicode 进行加载、或者设置用户剪贴板的内容。

在运行时获取有关用户系统的数据

通过检查 System.totalMemory 属性，可以确定 Flash Player 或 AIR 当前所用的内存数量（以字节为单位）。该属性可让您监视内存使用情况，并根据内存级别的更改方式优化应用程序。例如，如果特定视觉效果导致内存使用量大幅增加，您可能需要考虑修改此效果或将其完全消除。

`System.ime` 属性是对当前安装的输入法编辑器 (IME) 的引用。该属性允许使用 `addEventListener()` 方法来侦听 `imeComposition` 事件 (`flash.events.IMEEEvent.IME_COMPOSITION`)。

`System` 类中的第三个属性是 `useCodePage`。如果将 `useCodePage` 设置为 `true`, 则 Flash Player 和 AIR 会使用运行播放器的操作系统的传统代码页来加载外部文本文件。如果将此属性设置为 `false`, 则 Flash Player 或 AIR 按 Unicode 解释外部文件。

如果将 `System.useCodePage` 设置为 `true`, 请记住, 运行播放器的操作系统的传统代码页中必须包括在外部文本文件中使用的字符, 这样才能显示文本。例如, 如果您加载了一个包含中文字符的外部文本文件, 则这些字符不能显示在使用英文 Windows 代码页的系统上, 因为该代码页不包括中文字符。

要确保所有平台上的用户都能查看 SWF 文件中使用的外部文本文件, 应将所有外部文本文件按 `Unicode` 进行编码, 并将 `System.useCodePage` 设置保留为默认设置 `false`。这样, Flash Player 和 AIR 会按 `Unicode` 解释文本。

将文本保存到剪贴板

`System` 类包含一个名为 `setClipboard()` 的方法, 该方法使 Flash Player 和 AIR 可以使用指定的字符串来设置用户剪贴板的内容。出于安全方面的考虑, 不存在 `Security.getClipboard()` 方法, 因为这样的方法可能允许恶意站点访问最近复制到用户剪贴板中的数据。

下面的代码演示出现安全错误时如何将错误消息复制到用户剪贴板。如果用户要报告应用程序的潜在错误, 则错误消息会很有用。

```
private function securityErrorHandler(event:SecurityErrorEvent):void
{
    var errorString:String = "[" + event.type + "] " + event.text;
    trace(errorString);
    System.setClipboard(errorString);
}
```

使用 Capabilities 类

开发人员可通过 `Capabilities` 类来确定正在运行 SWF 文件的环境。使用 `Capabilities` 类的各种属性, 可以查明用户的分辨率、用户的系统是否支持辅助功能软件、用户操作系统的语言以及当前安装的 Flash Player 或 AIR 的版本。

通过检查 `Capabilities` 类的属性, 可以自定义应用程序, 使其以最佳方式与特定用户环境配合工作。例如, 通过检查 `Capabilities.screenResolutionX` 和 `Capabilities.screenResolutionY` 属性, 可以确定用户系统所使用的显示分辨率, 并确定最合适视频大小。或者, 在尝试加载外部 mp3 文件之前, 可以检查 `Capabilities.hasMP3` 属性以查看用户系统是否支持 mp3 播放。

以下代码使用正则表达式分析客户端所使用的 Flash Player 版本:

```
var versionString:String = Capabilities.version;
var pattern:RegExp = /^(\w*) (\d*),(\d*),(\d*),(\d*)$/;
var result:Object = pattern.exec(versionString);
if (result != null)
{
    trace("input: " + result.input);
    trace("platform: " + result[1]);
    trace("majorVersion: " + result[2]);
    trace("minorVersion: " + result[3]);
    trace("buildNumber: " + result[4]);
    trace("internalBuildNumber: " + result[5]);
}
else
{
    trace("Unable to match RegExp.");
}
```

如果要向某个服务器端脚本发送用户的系统功能, 以便将信息存储在数据库中, 则可以使用下面的 ActionScript 代码:

```
var url:String = "log_visitor.cfm";
var request:URLRequest = new URLRequest(url);
request.method = URLRequestMethod.POST;
request.data = new URLVariables(Capabilities.serverString);
var loader:URLLoader = new URLLoader(request);
```

使用 ApplicationDomain 类

ApplicationDomain 类的用途是存储 ActionScript 3.0 定义表。SWF 文件中的所有代码被定义为存在于应用程序域中。可以使用应用程序域划分位于同一个安全域中的类。这允许同一个类存在多个定义，并且还允许子级重用父级定义。

在使用 Loader 类 API 加载用 ActionScript 3.0 编写的外部 SWF 文件时，可以使用应用程序域。（请注意，在加载图像或用 ActionScript 1.0 或 ActionScript 2.0 编写的 SWF 文件时不能使用应用程序域。）包含在已加载类中的所有 ActionScript 3.0 定义都存储在应用程序域中。加载 SWF 文件时，通过将 LoaderContext 对象的 applicationDomain 参数设置为 ApplicationDomain.currentDomain，可以指定文件包含在 Loader 对象所在的相同应用程序域中。通过将加载的 SWF 文件放在同一个应用程序域中，可以直接访问它的类。如果加载的 SWF 文件包含嵌入的媒体（可通过其关联的类名称访问），或者您要访问加载的 SWF 文件的方法，则这种方式会很有用，如下面的示例所示：


```
package
{
    import flash.display.Loader;
    import flash.display.Sprite;
    import flash.events.*;
    import flash.net.URLRequest;
    import flash.system.ApplicationDomain;
    import flash.system.LoaderContext;

    public class ApplicationDomainExample extends Sprite
    {
        private var ldr:Loader;
        public function ApplicationDomainExample()
        {
            ldr = new Loader();
            var req:URLRequest = new URLRequest("Greeter.swf");
            var ldrContext:LoaderContext = new LoaderContext(false, ApplicationDomain.currentDomain);
            ldr.contentLoaderInfo.addEventListener(Event.COMPLETE, completeHandler);
            ldr.load(req, ldrContext);
        }
        private function completeHandler(event:Event):void
        {
            ApplicationDomain.currentDomain.getDefinition("Greeter");
            var myGreeter:Greeter = Greeter(event.target.content);
            var message:String = myGreeter.welcome("Tommy");
            trace(message); // Hello, Tommy
        }
    }
}
```

使用应用程序域时，还要记住以下几点：

- SWF 文件中的所有代码被定义为存在于应用程序域中。主应用程序在“当前域”中运行。“系统域”中包含所有应用程序域（包括当前域），也就是包含所有 Flash Player 类。
- 所有应用程序域（除系统域外）都有关联的父域。主应用程序的应用程序域的父域是系统域。已加载的类仅在其父级中没有相关定义时才进行定义。不能用较新的定义覆盖已加载类的定义。

下图显示了某个应用程序在单个域 (domain1.com) 中加载多个 SWF 文件的内容。根据加载内容的不同，可以使用不同的应用程序域。紧跟的文本说明用于为应用程序中的每个 SWF 文件设置适当应用程序域的逻辑。



A. 用法 A B. 用法 B C. 用法 C

主应用程序文件为 application1.swf。它包含从其它 SWF 文件加载内容的 Loader 对象。在此方案下，当前域为 Application domain 1。用法 A、用法 B 和用法 C 说明了为应用程序中的每个 SWF 文件设置适当应用程序域的不同方法。

用法 A 通过创建系统域的子级划分子级 SWF 文件。在示意图中，Application domain 2 创建为系统域的子级。application2.swf 文件在 Application domain 2 中加载，因此其类定义从 application1.swf 中定义的类中划分出来。

此方法的一个用处是使旧版应用程序能够动态加载相同应用程序的更新版本，而不会发生冲突。之所以不发生冲突，是因为尽管使用的是同样的类名称，但它们划分到不同的应用程序域中。

下面的代码创建作为系统域子级的一个应用程序域，并使用该应用程序域开始加载一个 SWF：

```
var appDomainA:ApplicationDomain = new ApplicationDomain();

var contextA:LoaderContext = new LoaderContext(false, appDomainA);
var loaderA:Loader = new Loader();
loaderA.load(new URLRequest("application2.swf"), contextA);
```

用法 B: 在当前类定义中添加新的类定义。module1.swf 的应用程序域设置为当前域 (Application domain 1)。这可让您将新的类定义添加到应用程序当前的一组类定义中。这可用于主应用程序的运行时共享库。加载的 SWF 被视为远程共享库 (RSL)。使用此方法可以在应用程序启动之前使用预加载器加载 RSL。

下面的代码加载一个 SWF，同时将其应用程序域设置为当前域：

```
var appDomainB:ApplicationDomain = ApplicationDomain.currentDomain;

var contextB:LoaderContext = new LoaderContext(false, appDomainB);
var loaderB:Loader = new Loader();
loaderB.load(new URLRequest("module1.swf"), contextB);
```

用法 C: 通过创建当前域的新子域，使用父级的类定义。module3.swf 的应用程序域是当前域的子级，并且子级使用所有类的父级的版本。此方法的一个用处可能是作为一个使用主应用程序的类型的多屏幕丰富 Internet 应用程序 (RIA) 模块，该模块作为主应用程序的子级加载。如果能够确保所有类始终更新为向后兼容，并且正在加载的应用程序始终比其加载的软件的版本新，则子级将使用父级版本。如果可以确保不继续拥有对子级 SWF 的引用，则拥有了新的应用程序域还使您能够卸载所有的类定义以便于垃圾回收。

此方法使加载的模块可以共享加载者的 singleton 对象和静态类成员。

下面的代码创建当前域的一个新子域，并使用该应用程序域开始加载一个 SWF：

```
var appDomainC:ApplicationDomain = new ApplicationDomain(ApplicationDomain.currentDomain);

var contextC:LoaderContext = new LoaderContext(false, appDomainC);
var loaderC:Loader = new Loader();
loaderC.load(new URLRequest("module3.swf"), contextC);
```

使用 IME 类

通过使用 IME 类，您可以在 Flash Player 或 Adobe AIR 中运用操作系统的 IME。

使用 ActionScript 可以确定以下内容：

- 用户的计算机上是否安装了 IME (Capabilities.hasIME)
- 用户计算机上是否启用了 IME (IME.enabled)
- 当前 IME 使用的转换模式 (IME.conversionMode)

可以使用特定 IME 上下文关联输入文本字段。在输入字段之间切换时，还可以在平假名（日文）、全角数字、半角数字、直接输入等之间切换 IME。

利用 IME，用户可键入多字节语言（例如中文、日文和韩文）的非 ASCII 文本字符。

有关使用 IME 的详细信息，请参阅要为其开发应用程序的操作系统的文档。要获取其它资源，请参阅以下 Web 站点：

- <http://www.microsoft.com/goglobal/>
- <http://developer.apple.com/documentation/>
- <http://www.java.sun.com/>

注：如果用户计算机上 IME 未处于活动状态，则对 IME 方法或属性（除 Capabilities.hasIME 之外）的调用将会失败。一旦手动激活 IME，随后对 IME 方法和属性的 ActionScript 调用即会正常运行。例如，如果使用日文 IME，则必须在调用任何 IME 方法或属性之前将它激活。

查看是否已安装并启用了 IME

调用任何 IME 方法或属性之前，总应检查用户计算机上当前是否已安装并启用 IME。以下代码说明了在调用任何方法之前如何检查用户是否安装并启用了 IME：

```
if (Capabilities.hasIME)
{
    if (IME.enabled)
    {
        trace("IME is installed and enabled.");
    }
    else
    {
        trace("IME is installed but not enabled. Please enable your IME and try again.");
    }
}
else
{
    trace("IME is not installed. Please install an IME and try again.");
}
```

上面的代码首先使用 Capabilities.hasIME 属性检查用户是否安装了 IME。如果此属性设置为 true，则代码使用 IME.enabled 属性检查当前是否启用了用户的 IME。

确定当前启用的是哪种 IME 转换模式

构建多语言应用程序时，您可能需要确定用户当前启用的是哪种 IME 转换模式。以下代码说明了如何检查用户是否安装了 IME，以及在安装 IME 的情况下当前启用的是哪种 IME 转换模式：

```
if (Capabilities.hasIME)
{
    switch (IME.conversionMode)
    {
        case IMEConversionMode.ALPHANUMERIC_FULL:
            tf.text = "Current conversion mode is alphanumeric (full-width).";
            break;
        case IMEConversionMode.ALPHANUMERIC_HALF:
            tf.text = "Current conversion mode is alphanumeric (half-width).";
            break;
        case IMEConversionMode.CHINESE:
            tf.text = "Current conversion mode is Chinese.";
            break;
        case IMEConversionMode.JAPANESE_HIRAGANA:
            tf.text = "Current conversion mode is Japanese Hiragana.";
            break;
        case IMEConversionMode.JAPANESE_KATAKANA_FULL:
            tf.text = "Current conversion mode is Japanese Katakana (full-width).";
            break;
        case IMEConversionMode.JAPANESE_KATAKANA_HALF:
            tf.text = "Current conversion mode is Japanese Katakana (half-width).";
            break;
        case IMEConversionMode.KOREAN:
            tf.text = "Current conversion mode is Korean.";
            break;
        default:
            tf.text = "Current conversion mode is " + IME.conversionMode + ".";
            break;
    }
}
else
{
    tf.text = "Please install an IME and try again.";
}
```

上面的代码首先检查用户是否安装了 IME。接下来，该代码通过将 IME.conversionMode 属性与 IMEConversionMode 类中的每个常量进行比较，检查当前 IME 使用的是哪种转换模式。

设置 IME 转换模式

更改用户的 IME 的转换模式时，您需要确保将代码封装在 try..catch 块中，因为使用 conversionMode 属性设置转换模式时，如果 IME 无法设置该转换模式，则可能会引发错误。下面的代码演示如何使用 try..catch 块（在设置 IME.conversionMode 属性时）：

```
var statusText:TextField = new TextField;
statusText.autoSize = TextFieldAutoSize.LEFT;
addChild(statusText);
if (Capabilities.hasIME)
{
    try
    {
        IME.enabled = true;
        IME.conversionMode = IMEConversionMode.KOREAN;
        statusText.text = "Conversion mode is " + IME.conversionMode + ".";
    }
    catch (error:Error)
    {
        statusText.text = "Unable to set conversion mode.\n" + error.message;
    }
}
```

上面的代码先创建一个文本字段，该字段用于向用户显示状态消息。接下来，如果已安装 IME，该代码会启用 IME 并将转换模式设置为“韩文”。如果用户计算机上未安装韩文 IME，则 Flash Player 或 AIR 将引发错误，该错误由 try..catch 块捕获。try..catch 块会在先前创建的文本字段中显示该错误消息。

为特定文本字段禁用 IME

在某些情况下，最好在用户键入字符时禁用用户的 IME。例如，如果有一个文本字段只接受数字输入，您可能不想让 IME 出现并减缓数据输入的速度。

下面的示例演示如何侦听 FocusEvent.FOCUS_IN 和 FocusEvent.FOCUS_OUT 事件并相应地禁用用户的 IME：

```
var phoneTxt:TextField = new TextField();
var nameTxt:TextField = new TextField();

phoneTxt.type = TextFieldType.INPUT;
phoneTxt.addEventListener(FocusEvent.FOCUS_IN, focusInHandler);
phoneTxt.addEventListener(FocusEvent.FOCUS_OUT, focusOutHandler);
phoneTxt.restrict = "0-9";
phoneTxt.width = 100;
phoneTxt.height = 18;
phoneTxt.background = true;
phoneTxt.border = true;
addChild(phoneTxt);

nameField.type = TextFieldType.INPUT;
nameField.x = 120;
nameField.width = 100;
nameField.height = 18;
nameField.background = true;
nameField.border = true;
addChild(nameField);

function focusInHandler(event:FocusEvent):void
{
    if (Capabilities.hasIME)
    {
        IME.enabled = false;
    }
}
function focusOutHandler(event:FocusEvent):void
{
    if (Capabilities.hasIME)
    {
        IME.enabled = true;
    }
}
```

此示例创建两个输入文本字段 phoneTxt 和 nameTxt，然后为 phoneTxt 文本字段添加两个事件侦听器。当用户将焦点设置为 phoneTxt 文本字段时，将调度 FocusEvent.FOCUS_IN 事件并禁用 IME。当 phoneTxt 文本字段失去焦点时，将调度 FocusEvent.FOCUS_OUT 事件以重新启用 IME。

侦听 IME 合成事件

设置合成字符串时会调度 IME 合成事件。例如，如果用户启用了 IME 并键入日文字符串，在用户选择合成字符串时，即会调度 IMEEvent.IME_COMPOSITION 事件。为了侦听 IMEEvent.IME_COMPOSITION 事件，您需要在 System 类的静态 ime 属性中添加一个事件侦听器 (flash.system.System.ime.addEventListener(...))，如下面的示例所示：

```
var inputTxt:TextField;
var outputTxt:TextField;

inputTxt = new TextField();
inputTxt.type = TextFieldType.INPUT;
inputTxt.width = 200;
inputTxt.height = 18;
inputTxt.border = true;
inputTxt.background = true;
addChild(inputTxt);

outputTxt = new TextField();
outputTxt.autoSize = TextFieldAutoSize.LEFT;
outputTxt.y = 20;
addChild(outputTxt);

if (Capabilities.hasIME)
{
    IME.enabled = true;
    try
    {
        IME.conversionMode = IMEConversionMode.JAPANESE_HIRAGANA;
    }
    catch (error:Error)
    {
        outputTxt.text = "Unable to change IME.";
    }
    System.ime.addEventListener(IMEEEvent.IME_COMPOSITION, imeCompositionHandler);
}
else
{
    outputTxt.text = "Please install IME and try again.";
}

function imeCompositionHandler(event:IMEEEvent):void
{
    outputTxt.text = "you typed: " + event.text;
}
```

上面的代码创建两个文本字段，并将其添加到显示列表中。第一个文本字段 `inputTxt` 是一个输入文本字段，用户可以在其中输入日文文本。第二个文本字段 `outputTxt` 是一个动态文本字段，用于向用户显示错误消息，或回显用户在 `inputTxt` 文本字段中键入的日文字符串。

示例：检测系统功能

`CapabilitiesExplorer` 示例演示如何使用 `flash.system.Capabilities` 类来确定用户的 Flash Player 或 AIR 版本支持哪些功能。该示例讲授以下方法：

- 使用 `Capabilities` 类检测用户的 Flash Player 或 AIR 版本支持哪些功能
- 使用 `ExternalInterface` 类检测用户的浏览器支持哪些浏览器设置

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。可以在 Samples/CapabilitiesExplorer 文件夹中找到 `CapabilitiesExplorer` 应用程序文件。该应用程序包含下列文件：

文件	描述
CapabilitiesExplorer.fla 或 CapabilitiesExplorer.mxml	Flash 或 Flex 中的主应用程序文件（分别为 FLA 和 MXML）。
com/example/programmingas3/capabilities/CapabilitiesGrabber.as	用于提供应用程序的主要功能的类，这些主要功能包括在数组中添加系统功能、对项目进行排序以及使用 ExternalInterface 类检索浏览器功能。
capabilities.html	包含与外部 API 进行通信所必需的 JavaScript 的 HTML 容器。

CapabilitiesExplorer 概述

CapabilitiesExplorer.mxml 文件负责设置 CapabilitiesExplorer 应用程序的用户界面。用户的 Flash Player 或 AIR 版本的功能将显示在舞台上的 DataGrid 组件实例中。如果用户是从 HTML 容器运行应用程序且外部 API 可用，也将显示其浏览器功能。

调度主应用程序文件的 creationComplete 事件时，将调用 initApp() 方法。initApp() 方法从 com.example.programmingas3.capabilities.CapabilitiesGrabber 类中调用 getCapabilities() 方法。initApp() 方法的代码如下所示：

```
private function initApp():void
{
    var dp:Array = CapabilitiesGrabber.getCapabilities();
    capabilitiesGrid.dataProvider = dp;
}
```

CapabilitiesGrabber.getCapabilities() 方法返回排序的 AIR 或 Flash Player 和浏览器功能的数组，然后将其设置为舞台上的 capabilitiesGrid DataGrid 组件实例的 dataProvider 属性。

CapabilitiesGrabber 类概述

CapabilitiesGrabber 类的静态 getCapabilities() 方法将 flash.system.Capabilities 类中的每个属性添加到数组 (capDP)。然后调用 CapabilitiesGrabber 类中的静态 getBrowserObjects() 方法。getBrowserObjects() 方法使用外部 API 循环访问浏览器的导航器对象（包含浏览器的功能）。getCapabilities() 方法如下所示：

```
public static function getCapabilities():Array
{
    var capDP:Array = new Array();
    capDP.push({name:"Capabilities.avHardwareDisable", value:Capabilities.avHardwareDisable});
    capDP.push({name:"Capabilities.hasAccessibility", value:Capabilities.hasAccessibility});
    capDP.push({name:"Capabilities.hasAudio", value:Capabilities.hasAudio});
    ...
    capDP.push({name:"Capabilities.version", value:Capabilities.version});
    var navArr:Array = CapabilitiesGrabber.getBrowserObjects();
    if (navArr.length > 0)
    {
        capDP = capDP.concat(navArr);
    }
    capDP.sortOn("name", Array.CASEINSENSITIVE);
    return capDP;
}
```

getBrowserObjects() 方法返回一个数组，该数组包含浏览器的导航器对象中的每个属性。如果该数组的长度为一个或多个项目，则将浏览器功能的数组 (navArr) 追加到 Flash Player 功能的数组 (capDP) 末尾，并按字母顺序对整个数组进行排序。最后，排序后的数组将返回到主应用程序文件，随后填充数据网格。getBrowserObjects() 方法的代码如下所示：

```
private static function getBrowserObjects():Array
{
    var itemArr:Array = new Array();
    var itemVars:URLVariables;
    if (ExternalInterface.available)
    {
        try
        {
            var tempStr:String = ExternalInterface.call("JS_getBrowserObjects");
            itemVars = new URLVariables(tempStr);
            for (var i:String in itemVars)
            {
                itemArr.push({name:i, value:itemVars[i]});
            }
        }
        catch (error:SecurityError)
        {
            // ignore
        }
    }
    return itemArr;
}
```

如果当前用户环境中外部 API，则 Flash Player 将调用 JavaScript `JS_getBrowserObjects()` 方法，该方法循环访问浏览器的导航器对象并向 ActionScript 返回一个 URL 编码的值的字符串。该字符串随后将转换为 `URLVariables` 对象 (`itemVars`) 并添加到 `itemArr` 数组中，将向调用脚本返回该数组。

与 JavaScript 进行通信

作为构建 CapabilitiesExplorer 应用程序的最后一部分工作，要编写循环访问浏览器导航器对象中的每个项目所必需的 JavaScript，并将一个名称 / 值对追加到一个临时数组的末尾。container.html 文件中的 JavaScript JS_getBrowserObjects() 方法的代码如下所示：

```
}

// Loop over each item in the browser's screen object.
for (var name in screen)
{
    var value = screen[name];

    // If the current value is a number, add it to the array, otherwise
    // ignore the item.
    switch (typeof(value))
    {
        case "number":
            var tempStr = "screen." + name + "=" + escape(value);
            tempArr.push(tempStr);
            break;
    }
}
// Return the array as a URL-encoded string of name-value pairs.
return tempArr.join("&");

</script>
```

代码首先创建一个临时数组，该数组用于保存导航器对象中的所有名称 / 值对。接下来，使用 `for..in` 循环对导航器对象进行循环访问，并计算出当前值的数据类型以过滤掉不需要的值。在此应用程序中，我们只需要 `String` 或 `Boolean` 值，其它数据类型（例如函数或数组）将被忽略。导航器对象中的每个 `String` 或 `Boolean` 值将追加到 `tempArr` 数组末尾。然后，使用 `for..in` 循环对浏览器的屏幕对象进行循环访问，并将每个数值添加到 `tempArr` 数组中。最后，使用 `Array.join()` 方法将临时数组转换为字符串。该数组使用 `(&)` 符号作为分隔符，这使得 ActionScript 可以使用 `URLVariables` 类轻松分析数据。

第 29 章：复制和粘贴

使用剪贴板 API 中的类将信息复制到系统剪贴板中或从中复制信息。可以与在 Adobe® AIR™ 和 Adobe® Flash® Player 中运行的应用程序实现数据传输的格式包括：

- 位图（仅限 AIR）
- 文件（仅限 AIR）
- 文本
- HTML 格式的文本
- RTF 数据
- URL 字符串（仅限 AIR）
- 序列化对象
- 对象引用（仅在原始应用程序内有效）

复制和粘贴基础知识

复制和粘贴 API 包含以下类。

包	类
flash.desktop	<ul style="list-style-type: none"> • Clipboard • ClipboardFormats • ClipboardTransferMode

静态 Clipboard.generalClipboard 属性表示操作系统剪贴板。Clipboard 类为从 Clipboard 对象读取数据或向其中写入数据提供了方法。

HTMLLoader 类（在 AIR 中）和 TextField 类用于实现一般复制和粘贴快捷键的默认行为。若要实现自定义组件的复制和粘贴快捷键行为，您可以直接侦听这些键击。您也可以使用本机菜单命令及等效键来间接响应键击。

可以在一个 Clipboard 对象中包含同一信息的不同表示形式，以使其它应用程序更易于理解和使用其中的数据。例如，图像可以以图像数据形式、序列化的 Bitmap 对象形式和文件形式包含在其中。以某种格式呈现数据的操作可以延迟，以便直到读取此格式的数据时才真正创建此格式。

读取和写入系统剪贴板

若要读取操作系统剪贴板，请调用 Clipboard.generalClipboard 对象的 getData() 方法，并传递要读取的格式的名称：

```
import flash.desktop.Clipboard;
import flash.desktop.ClipboardFormats;

if(Clipboard.generalClipboard.hasFormat(ClipboardFormats.TEXT_FORMAT)){
    var text:String = Clipboard.generalClipboard.getData(ClipboardFormats.TEXT_FORMAT);
}
```

注 在 Flash Player 中运行的内容或在 AIR 的非应用程序沙箱中运行的内容只能在 `paste` 事件的事件处理函数中调用 `getData()` 方法。只有在 AIR 应用程序沙箱中运行的代码才能在 `paste` 事件处理函数的外部调用 `getData()` 方法。

若要写入剪贴板，请以一种或多种格式将数据添加到 `Clipboard.generalClipboard` 对象。任何同一格式的现有数据都将被自动覆盖。然而，建议在将新数据写入系统剪贴板之前清除系统剪贴板，这样可确保任何其它格式的无关数据也将删除。

```
import flash.desktop.Clipboard;
import flash.desktop.ClipboardFormats;

var textToCopy:String = "Copy to clipboard.";
Clipboard.generalClipboard.clear();
Clipboard.generalClipboard.setData(ClipboardFormats.TEXT_FORMAT, textToCopy, false);
```

注：在 Flash Player 中运行的内容或在 AIR 的非应用程序沙箱中运行的内容只能在用户事件（如键盘或鼠标事件，或者 `copy` 或 `cut` 事件）的事件处理函数中调用 `setData()` 方法。只有在 AIR 应用程序沙箱中运行的代码才能在用户事件处理函数的外部调用 `setData()` 方法。

剪贴板数据格式

剪贴板格式描述 `Clipboard` 对象中放置的数据。Flash Player 或 AIR 可在 ActionScript 数据类型和系统剪贴板格式之间自动进行标准数据格式的转换。此外，使用应用程序定义的格式可以在基于 ActionScript 的应用程序内或这些应用程序之间传输应用程序对象。

`Clipboard` 对象可以包含采用不同格式的同一信息的多种表示形式。例如，表示 `Sprite` 的 `Clipboard` 对象可以包括在同一应用程序中使用的引用格式、由在 Flash Player 或 AIR 中运行的另一应用程序使用的序列化格式、由图像编辑器使用的位图格式以及文件列表格式，还可能具有延迟呈现功能以对 PNG 文件进行编码，以便将 `Sprite` 的表示形式复制或拖动到文件系统。

标准数据格式

`ClipboardFormats` 类中提供了用于定义标准格式名称的常量：

常量	描述
TEXT_FORMAT	文本格式数据与 ActionScript <code>String</code> 类之间的转换。
HTML_FORMAT	带有 HTML 标记的文本。
RICH_TEXT_FORMAT	RTF 格式数据与 ActionScript <code>ByteArray</code> 类之间的转换。不会以任何方式对 RTF 标记进行解释或转换。
BITMAP_FORMAT	(仅限 AIR) 位图格式数据与 ActionScript <code>BitmapData</code> 类之间的转换。
FILE_LIST_FORMAT	(仅限 AIR) 文件列表格式数据与 ActionScript <code>File</code> 对象数组之间的转换。
URL_FORMAT	(仅限 AIR) URL 格式数据与 ActionScript <code>String</code> 类之间的转换。

自定义数据格式

您可以使用应用程序定义的自定义格式将对象作为引用或序列化副本传输。首选参数仅在 AIR 或 Flash Player 的同一应用程序中运行时才有效。序列化对象可以在运行于 AIR 或 Flash Player 中的应用程序之间传输，但只能用于序列化和反序列化后仍然有效的对象。如果对象的属性为简单类型或可序列化对象，则通常可以将该对象序列化。

若要将序列化对象添加到 `Clipboard` 对象，请在调用 `Clipboard.setData()` 方法时将可序列化参数设置为 `true`。格式名称可以是标准格式中的某一种或由应用程序定义的任意字符串。

传输模式

使用自定义数据格式将对象写入剪贴板后，可以从剪贴板中将对象数据作为引用读取，也可以将其作为原始对象的序列化副本读取。AIR 定义了四种传输模式，这些模式确定了对象是以引用还是以序列化副本的形式传输：

传输模式	描述
ClipboardTransferModes.ORIGINAL_ONLY	仅返回引用。如果没有引用，则返回 null 值。
ClipboardTransferModes.ORIGINAL_PREFERRED	如果存在引用，则返回引用。否则返回序列化副本。
ClipboardTransferModes.CLONE_ONLY	仅返回序列化副本。如果没有可用的序列化副本，则返回 null 值。
ClipboardTransferModes.CLONE_PREFERRED	如果存在序列化副本，则返回序列化副本。否则返回引用。

读取和写入自定义数据格式

将对象写入剪贴板时，可以将任何不以保留前缀 air: 或 flash: 开头的字符串用于格式参数。请使用相同字符串作为读取对象的格式。下面的示例说明如何从剪贴板读取对象和向其中写入对象：

```
public function createClipboardObject(object:Object):Clipboard{
    var transfer:Clipboard = Clipboard.generalClipboard;
    transfer.setData("object", object, true);
}
```

若要从 Clipboard 对象中提取序列化对象（放置或粘贴操作之后），请使用相同格式名称和 cloneOnly 或 clonePreferred 传输模式。

```
var transfer:Object = clipboard.getData("object", ClipboardTransferMode.CLONE_ONLY);
```

此时将始终向 Clipboard 对象添加一个引用。若要从 Clipboard 对象中提取引用（放置或粘贴操作之后）而不是提取序列化副本，请使用 originalOnly 或 originalPreferred 传输模式：

```
var transferredObject:Object =
    clipboard.getData("object", ClipboardTransferMode.ORIGINAL_ONLY);
```

仅当 Clipboard 对象源自自 AIR 或 Flash Player 中运行的当前应用程序时，引用才有效。当引用可用时，可使用 originalPreferred 传输模式访问该引用；当引用不可用时，则可使用该模式访问序列化副本。

延迟呈现

如果创建数据格式的计算成本高昂，则可以通过提供按需提供数据的函数来使用延迟呈现。仅当放置或粘贴操作的接收方请求延迟格式的数据时才会调用此函数。

呈现函数是通过使用 setDataHandler() 方法添加到 Clipboard 对象中的。该函数必须返回相应格式的数据。例如，如果您调用 setDataHandler(ClipboardFormat.TEXT_FORMAT, writeText)，则 writeText() 函数必须返回字符串。

如果使用 setData() 方法将同一类型的数据格式添加到 Clipboard 对象，则该数据的优先级高于其延迟版本（不会调用呈现函数）。如果再次访问该剪贴板数据，则可能会再次调用或不调用该呈现函数。

注：在 Mac OS X 上，延迟呈现仅适用于自定义数据格式。使用标准数据格式时，会直接调用呈现函数。

使用延迟呈现函数粘贴文本

下面的示例说明如何实现延迟呈现函数。

当用户按下“Copy”按钮时，应用程序将清除系统剪贴板以确保不会留下上次剪贴板操作的数据。然后，setDataHandler() 方法将 renderData() 函数设置为剪贴板渲染器。

当用户从目标文本字段的上下文菜单中选择“粘贴”命令时，应用程序将访问剪贴板并设置目标文本。由于已使用函数而不是字符串设置了剪贴板上的文本数据格式，剪贴板将调用 `renderData()` 函数。`renderData()` 函数将返回源文本中的文本，然后将其分配给目标文本。

请注意，如果按下“Paste”按钮前编辑源文本，则编辑操作将反映到粘贴的文本中，即使按下“Paste”按钮后再进行编辑也是如此。这是因为呈现函数直到按下“Paste”按钮后才会复制源文本。（当在实际的应用程序中使用延迟呈现时，最好以某种方式存储或保护源数据以防止出现此问题。）

```
package {
    import flash.desktop.Clipboard;
    import flash.desktop.ClipboardFormats;
    import flash.desktop.ClipboardTransferMode;
    import flash.display.Sprite;
    import flash.text.TextField;
    import flash.text.TextFormat;
    import flash.text.TextFieldType;
    import flash.events.MouseEvent;
    import flash.events.Event;
    public class DeferredRenderingExample extends Sprite
    {
        private var sourceTextField:TextField;
        private var destination:TextField;
        private var copyText:TextField;
        public function DeferredRenderingExample():void
        {
            sourceTextField = createTextField(10, 10, 380, 90);
            sourceTextField.text = "Neque porro quisquam est qui dolorem "
                + "ipsum quia dolor sit amet, consectetur, adipisci velit.";

            copyText = createTextField(10, 110, 35, 20);
            copyText.htmlText = "<a href='#!'>Copy</a>";
            copyText.addEventListener(MouseEvent.CLICK, onCopy);

            destination = createTextField(10, 145, 380, 90);
            destination.addEventListener(Event.PASTE, onPaste);
        }
        private function createTextField(x:Number, y:Number, width:Number,
            height:Number):TextField
        {
            var newTxt:TextField = new TextField();
            newTxt.x = x;
            newTxt.y = y;
            newTxt.height = height;
            newTxt.width = width;
            newTxt.border = true;
            newTxt.multiline = true;
            newTxt.wordWrap = true;
            newTxt.type = TextFieldType.INPUT;
            addChild(newTxt);
            return newTxt;
        }
        public function onCopy(event:MouseEvent):void
        {
            Clipboard.generalClipboard.clear();
            Clipboard.generalClipboard.setDataHandler(ClipboardFormats.TEXT_FORMAT,
                renderData);
        }
        public function onPaste(event:Event):void
        {
```

```
sourceTextField.text =
Clipboard.generalClipboard.getData(ClipboardFormats.TEXT_FORMAT).toString;
}

public function renderData():String
{
    trace("Rendering data");
    var sourceStr:String = sourceTextField.text;
    if (sourceTextField.selectionEndIndex >
        sourceTextField.selectionBeginIndex)
    {
        return sourceStr.substring(sourceTextField.selectionBeginIndex,
                                   sourceTextField.selectionEndIndex);
    }
    else
    {
        return sourceStr;
    }
}
}
```

第 30 章：打印

Adobe® Flash® Player 和 Adobe® AIR™ 可以与操作系统的打印接口通信，以便将页面传递给打印后台处理程序。Flash Player 或 AIR 发送到后台处理程序的每个页面都可能包含可见的、动态的或屏幕上未显示给用户的内容，其中包括数据库值和动态文本。另外，Flash Player 和 AIR 根据用户的打印机设置来设置 `flash.printing.PrintJob` 类的属性，以便您可以适当地设置页面格式。

本章详细介绍使用 `flash.printing.PrintJob` 类方法和属性创建打印作业、读取用户的打印设置以及根据 Flash Player 或 AIR 和用户操作系统的反馈调整打印作业的策略。

打印基础知识

打印简介

在 ActionScript 3.0 中，可以使用 `PrintJob` 类来创建显示内容的快照以转换为打印输出中的墨水和纸张表示形式。在某些方面，设置要打印的内容与设置在屏幕上显示的内容是相同的；即可以放置元素和调整其大小以创建所需的布局。但是，打印具有某些特性，而使其不同于屏幕布局。例如，打印机使用的分辨率不同于计算机显示器；计算机屏幕的内容是动态的并且可能会发生变化，而打印的内容本身静态的；在准备进行打印时，您需要考虑固定纸张大小的限制以及多页打印的可能性。

即使这些差异看起来是显而易见的，但在使用 ActionScript 设置打印时一定要记住这些不同之处。由于精确打印取决于您指定的值和用户打印机特性，因此，您可以使用 `PrintJob` 类中包含的属性来确定需要考虑的重要用户打印机特性。

常见打印任务

本章介绍了以下常见的打印任务：

- 启动打印作业
- 在打印作业中添加页面
- 确定用户是否取消了打印作业
- 指定是使用位图还是矢量呈现
- 设置页面大小、缩放和方向
- 指定可打印的内容区域
- 将屏幕大小转换为页面大小
- 打印多页打印作业

重要概念和术语

以下参考列表包含将会在本章中使用的重要术语：

- 后台处理程序：操作系统或打印机驱动程序软件的一部分，用于在等待打印页面时跟踪页面，并在打印机可用时将其发送到打印机。
- 页面方向：打印的内容相对于纸张的旋转角度，即水平（横向）或垂直（纵向）。
- 打印作业：组成单个打印输出的页面或页面集。

完成本章中的示例

学习本章的过程中，您可能想要测试示例代码清单。本章中的许多代码清单是较小的代码部分，而不是完整的打印工作示例或用于检查值的完整代码。测试这些示例涉及创建要打印的元素以及将代码清单与这些元素一起使用。本章中的最后两个示例是完整的打印示例；这些示例包含用于定义要打印的内容的代码，以及执行打印任务的代码。

要测试示例代码清单，请执行以下操作：

- 1 创建一个新的 Flash 文档。
- 2 选择时间轴的第 1 帧中的关键帧，并打开“动作”面板。
- 3 将代码清单复制到“脚本”窗格中。
- 4 从主菜单中，选择“控制”>“测试影片”以创建 SWF 文件并测试该示例。

打印页面

使用 PrintJob 类的实例来处理打印。要通过 Flash Player 或 AIR 打印基本页面，请依次使用下面四个语句：

- new PrintJob(): 创建指定的打印作业名称的新实例。
- PrintJob.start(): 为操作系统启动打印过程（系统将为用户调用打印对话框），并填充打印作业的只读属性。
- PrintJob.addPage(): 包含有关打印作业内容的详细信息，其中包括 Sprite 对象（及其包含的任何子级）、打印区域的大小以及打印机应将图像打印为矢量图形还是位图图像。您可以使用对 addPage() 的连续调用，在多个页面上打印多个 sprite。
- PrintJob.send(): 将页面发送到操作系统的打印机。

例如，一个非常简单的打印作业脚本如下所示（包括用于编译的 package、import 和 class 语句）：

```
package
{
    import flash.printing.PrintJob;
    import flash.display.Sprite;

    public class BasicPrintExample extends Sprite
    {
        var myPrintJob:PrintJob = new PrintJob();
        var mySprite:Sprite = new Sprite();

        public function BasicPrintExample()
        {
            myPrintJob.start();
            myPrintJob.addPage(mySprite);
            myPrintJob.send();
        }
    }
}
```

注：此示例是为了说明打印作业脚本的基本元素，不包含任何错误处理。要生成脚本以正确响应用户取消打印作业的操作，请参阅第 564 页的“[处理异常和返回值](#)”。

如果您出于任何原因而需要清除 PrintJob 对象的属性，请将 PrintJob 变量设置为 null（如 myPrintJob = null 中所示）。

Flash Player 和 AIR 任务和系统打印

由于 Flash Player 和 AIR 向操作系统的打印接口调度页面，因此您应了解 Flash Player 和 AIR 所管理的任务范围以及操作系统自身的打印接口所管理的任务范围。Flash Player 和 AIR 可启动打印作业、读取打印机的部分页面设置、将打印作业的内容传递给操作系统，并验证用户或系统是否已取消打印作业。其它过程（例如显示特定于打印机的对话框、取消后台打印作业或报告打印机的状态）都由操作系统处理。如果出现打印作业启动或格式设置问题，Flash Player 和 AIR 能够做出响应，但只能报告操作系统打印接口的某些属性或条件。作为开发人员，您编写的代码应能响应这些属性或条件。

处理异常和返回值

您应检查在用户已取消打印作业的情况下，在执行 `addPage()` 和 `send()` 调用之前 `PrintJob.start()` 方法是否返回 `true`。一种在继续之前检查是否已取消这些方法的简单途径是，将这些方法包含在 `if` 语句中，如下所示：

```
if (myPrintJob.start())
{
    // addPage() and send() statements here
}
```

如果 `PrintJob.start()` 为 `true`，即表示用户已选择 Print（或者 Flash Player 或 AIR 已启用 Print 命令），则可以调用 `addPage()` 和 `send()` 方法。

另外，为了帮助管理打印过程，Flash Player 和 AIR 对 `PrintJob.addPage()` 方法引发异常，以便您可以捕获错误并向用户提供信息和选项。如果 `PrintJob.addPage()` 方法失败，则可以调用其它函数或停止当前打印作业。通过将 `addPage()` 调用嵌入 `try..catch` 语句，可以捕获这些异常，如下面的示例所示。在该例中，`[params]` 是指定实际打印内容的参数的占位符：

```
if (myPrintJob.start())
{
    try
    {
        myPrintJob.addPage([params]);
    }
    catch (error:Error)
    {
        // Handle error,
    }
    myPrintJob.send();
}
```

打印作业启动后，您就可以使用 `PrintJob.addPage()` 添加内容，并查看是否生成异常（例如，如果用户已取消打印作业）。如果生成异常，则可以向 `catch` 语句添加逻辑，以便向用户（或者 Flash Player 或 AIR）提供信息和选项，也可以停止当前打印作业。如果成功添加了页面，则可以继续使用 `PrintJob.send()` 将页面发送到打印机。

如果 Flash Player 或 AIR 在将打印作业发送到打印机时遇到问题（例如，如果打印机处于脱机状态），也可以捕获该异常并向用户（或者 Flash Player 或 AIR）提供信息或更多选项（例如显示消息文本或在动画中提供警告）。例如，您可以为 `if..else` 语句中的文本字段分配新文本，如下面的代码所示：

```

if (myPrintJob.start())
{
    try
    {
        myPrintJob.addPage([params]);
    }
    catch (error:Error)
    {
        // Handle error.
    }
    myPrintJob.send();
}
else
{
    myAlert.text = "Print job canceled";
}

```

有关运行正常的示例，请参阅第 569 页的“[示例：缩放、裁剪和拼接](#)”。

处理页面属性

用户在“打印”对话框中单击“确定”且 PrintJob.start() 返回 true 后，您就可以访问由打印机设置定义的属性。这些设置包括纸张宽度、纸张高度（pageHeight 和 pageWidth）以及纸张上内容的方向。由于这些设置属于打印机设置，不受 Flash Player 或 AIR 控制，因此您无法更改这些设置；但是，您可以使用这些设置将发送到打印机的内容对齐以与当前设置匹配。有关详细信息，请参阅第 566 页的“[设置大小、缩放和方向](#)”。

设置矢量或位图呈现

您可以将打印作业手动设置为将每个页面在后台处理为矢量图形或位图图像。在某些情况下，矢量打印相对于位图打印而言会生成更小的后台文件和效果更佳的图像。但是，如果您的内容包含位图图像，并且您要保留任何 Alpha 透明度或色彩效果，则应将页面打印为位图图像。另外，非 PostScript 打印机会自动将任何矢量图形转换为位图图像。

注：Mac OS 上的 Adobe AIR 不支持矢量打印。

您可以在 PrintJob.addPage() 的第三个参数中指定位图打印，方法是传递一个 PrintJobOptions 对象并将 printAsBitmap 参数设置为 true，如下所示：

```

var options:PrintJobOptions = new PrintJobOptions();
options.printAsBitmap = true;
myPrintJob.addPage(mySprite, null, options);

```

如果没有指定第三个参数的值，打印作业将使用默认设置，即矢量打印。

注：如果您不想指定 printArea（第二个参数）的值，但想指定位图打印的值，请为 printArea 使用 null。

为打印作业语句定时

与 ActionScript 的先前版本不同，ActionScript 3.0 未将 PrintJob 对象限定在单帧。然而，由于在用户单击“打印”对话框中的“确定”按钮之后，操作系统会向用户显示打印状态信息，所以应尽快调用 PrintJob.addPage() 和 PrintJob.send()，以将页面发送到后台处理程序。如果到达包含 PrintJob.send() 调用的帧时发生延迟，将会延迟打印过程。

在 ActionScript 3.0 中，脚本超时限制为 15 秒。因此，打印作业序列中各个主要语句间的时间间隔不能超过 15 秒。也就是说，15 秒脚本超时限制适用于以下时间间隔：

- PrintJob.start() 和第一个 PrintJob.addPage() 之间
- PrintJob.addPage() 和下一个 PrintJob.addPage() 之间
- 最后一个 PrintJob.addPage() 和 PrintJob.send() 之间

如果以上任何一个间隔时间超过 15 秒，则对 PrintJob 实例的下一次 PrintJob.start() 调用会返回 false，并且下次对 PrintJob 实例执行 PrintJob.addPage() 时会导致 Flash Player 或 AIR 引发运行时异常。

设置大小、缩放和方向

第 563 页的“[打印页面](#)”一节详细介绍基本打印作业的步骤，在基本打印作业中，输出直接反映了指定的 Sprite 的屏幕大小和位置的打印效果。然而，打印机使用不同的分辨率进行打印，并且可以具有对打印 sprite 的外观有不利影响的设置。

Flash Player 和 AIR 可以读取操作系统的打印设置，但请注意，这些属性是只读的：虽然可以响应这些属性的值，但无法进行设置。例如，您可以查明打印机的页面大小设置，并调整内容以适合该大小。您还可以确定打印机的边距设置和页面方向。为响应打印机设置，您可能需要指定打印区域、调整屏幕的分辨率与打印机的磅度量单位之间的差异，或转换内容以符合用户打印机的大小或方向设置。

为打印区域使用矩形

使用 PrintJob.addPage() 方法可以指定要打印的 sprite 的区域。第二个参数 printArea 为 Rectangle 对象的形式。您可以选择三种方法来提供该参数的值：

- 创建一个具有特定属性的 Rectangle 对象，然后将该矩形用于 addPage() 调用，如下例所示：

```
private var rect1:Rectangle = new Rectangle(0, 0, 400, 200);  
myPrintJob.addPage(sheet, rect1);
```

- 如果您尚未指定 Rectangle 对象，则可以在该调用本身内指定，如下面的示例所示：

```
myPrintJob.addPage(sheet, new Rectangle(0, 0, 100, 100));
```

- 如果您打算为 addPage() 调用中的第三个参数提供值，但不想指定矩形，则可以对第二个参数使用 null，如下所示：

```
myPrintJob.addPage(sheet, null, options);
```

注：如果您打算为打印尺寸指定矩形，请记住要导入 flash.display.Rectangle 类。

比较磅和像素

矩形的宽度和高度以像素为单位。打印机使用磅来作为打印的度量单位。磅的实际大小是固定的（1/72 英寸），但是在屏幕上，像素的大小取决于特定屏幕的分辨率。像素和磅之间的转换比率取决于打印机设置以及 sprite 是否经过缩放。一个 72 个像素宽的 sprite 在未经缩放的情况下打印输出将为一英寸宽，这时，一磅等于一个像素，且与屏幕分辨率无关。

您可以使用以下换算公式将英寸或厘米转换为缇或磅（1 缇为 1/20 磅）：

- 1 磅 = 1/72 英寸 = 20 缇
- 1 英寸 = 72 磅 = 1440 缇
- 1 厘米 = 567 缇

如果省略了 printArea 参数或错误地传递了该参数，将打印 sprite 的整个区域。

缩放

如果要在打印前对 Sprite 对象进行缩放，请在调用 PrintJob.addPage() 方法之前设置缩放属性（请参阅第 246 页的“[处理大小和缩放对象](#)”），并在打印后将这些属性设置回原始值。Sprite 对象的缩放与 printArea 属性无关。也就是说，如果指定一个 50 x 50 个像素的打印区域，则会打印 2500 个像素。如果对 Sprite 对象进行缩放，则同样会打印 2500 个像素，但按缩放后的大小打印 Sprite 对象。

有关示例，请参阅第 569 页的“[示例：缩放、裁剪和拼接](#)”。

横向或纵向打印

由于 Flash Player 和 AIR 可以检测方向设置，因此，您可以在 ActionScript 中构建逻辑来调整内容大小或旋转以响应打印机设置，如下面的示例所示：

```
if (myPrintJob.orientation == PrintJobOrientation.LANDSCAPE)
{
    mySprite.rotation = 90;
}
```

注：如果您计划读取纸张上内容方向的系统设置，请记住要导入 [PrintJobOrientation 类](#)。PrintJobOrientation 类提供了定义页面上的内容方向的常数值。使用以下语句导入该类：

```
import flash.printing.PrintJobOrientation;
```

响应页面高度和宽度

使用类似于处理打印机方向设置的策略，可以读取页面高度和宽度设置，并通过将某些逻辑嵌入 if 语句中来响应这些设置。下面的代码显示了一个示例：

```
if (mySprite.height > myPrintJob.pageHeight)
{
    mySprite.scaleY = .75;
}
```

此外，还可以通过比较页面尺寸和纸张尺寸来确定页面的边距设置，如下面的示例所示：

```
margin_height = (myPrintJob.paperHeight - myPrintJob.pageHeight) / 2;
margin_width = (myPrintJob.paperWidth - myPrintJob.pageWidth) / 2;
```

示例：多页打印

打印多页内容时，可以将每一页内容与不同的 sprite（本例中为 sheet1 和 sheet2）关联，然后为每个 sprite 使用 `PrintJob.addPage()`。以下代码说明了这种方法：

```
package
{
    import flash.display.MovieClip;
    import flash.printing.PrintJob;
    import flash.printing.PrintJobOrientation;
    import flash.display.Stage;
    import flash.display.Sprite;
    import flash.text.TextField;
    import flash.geom.Rectangle;

    public class PrintMultiplePages extends MovieClip
    {
        private var sheet1:Sprite;
        private var sheet2:Sprite;

        public function PrintMultiplePages():void
        {
            init();
            printPages();
        }

        private function init():void
        {
            sheet1 = new Sprite();
            createSheet(sheet1, "Once upon a time...", {x:10, y:50, width:80, height:130});
        }
    }
}
```

```
sheet2 = new Sprite();
createSheet(sheet2, "There was a great story to tell, and it ended quickly.\n\nThe end.", null);
}

private function createSheet(sheet:Sprite, str:String, imgValue:Object):void
{
    sheet.graphics.beginFill(0xEEEEEE);
    sheet.graphics.lineStyle(1, 0x000000);
    sheet.graphics.drawRect(0, 0, 100, 200);
    sheet.graphics.endFill();

    var txt:TextField = new TextField();
    txt.height = 200;
    txt.width = 100;
    txt.wordWrap = true;
    txt.text = str;

    if (imgValue != null)
    {
        var img:Sprite = new Sprite();
        img.graphics.beginFill(0xFFFFFFFF);
        img.graphics.drawRect(imgValue.x, imgValue.y, imgValue.width, imgValue.height);
        img.graphics.endFill();
        sheet.addChild(img);
    }
    sheet.addChild(txt);
}

private function printPages():void
{
    var pj:PrintJob = new PrintJob();
    var pagesToPrint:uint = 0;
    if (pj.start())
    {
        if (pj.orientation == PrintJobOrientation.LANDSCAPE)
        {
            throw new Error("Page is not set to an orientation of portrait.");
        }

        sheet1.height = pj.pageHeight;
        sheet1.width = pj.pageWidth;
        sheet2.height = pj.pageHeight;
        sheet2.width = pj.pageWidth;

        try
        {
            pj.addPage(sheet1);
            pagesToPrint++;
        }
        catch (error:Error)
        {
            // Respond to error.
        }
    }
}
```

```
        }

        try
        {
            pj.addPage(sheet2);
            pagesToPrint++;
        }
        catch (error:Error)
        {
            // Respond to error.
        }

        if (pagesToPrint > 0)
        {
            pj.send();
        }
    }
}
}
```

示例：缩放、裁剪和拼接

在某些情况下，为消除在屏幕上与打印纸张上的显示差异，打印显示对象时您可能需要调整显示对象的大小（或其它属性）。在打印之前调整显示对象的属性（例如使用 `scaleX` 和 `scaleY` 属性来调整）时，请注意，如果对象缩放的尺度超出了为打印区域定义的矩形，该对象将被裁剪。页面打印完毕后，您可能还需要重置属性。

以下代码对 `txt` 显示对象的尺寸进行缩放（但不缩放绿色背景框），结果将按照指定矩形的尺寸对文本字段进行裁剪。打印后，文本字段将返回到在屏幕上显示的原始大小。如果用户从操作系统的“打印”对话框取消打印作业，则 Flash Player 或 AIR 中的内容将更改以警告用户作业已取消。

```
package
{
    import flash.printing.PrintJob;
    import flash.display.Sprite;
    import flash.text.TextField;
    import flash.display.Stage;
    import flash.geom.Rectangle;

    public class PrintScaleExample extends Sprite
    {
        private var bg:Sprite;
        private var txt:TextField;

        public function PrintScaleExample():void
        {
            init();
            draw();
            printPage();
        }

        private function printPage():void
        {
            var pj:PrintJob = new PrintJob();
            txt.scaleX = 3;
            txt.scaleY = 2;
            if (pj.start())
            {
                trace(">> pj.orientation: " + pj.orientation);
```

```
trace(">> pj.pageWidth: " + pj.pageWidth);
trace(">> pj.pageHeight: " + pj.pageHeight);
trace(">> pj.paperWidth: " + pj.paperWidth);
trace(">> pj.paperHeight: " + pj.paperHeight);

try
{
    pj.addPage(this, new Rectangle(0, 0, 100, 100));
}
catch (error:Error)
{
    // Do nothing.
}
pj.send();
}
else
{
    txt.text = "Print job canceled";
}
// Reset the txt scale properties.
txt.scaleX = 1;
txt.scaleY = 1;
}

private function init():void
{
    bg = new Sprite();
    bg.graphics.beginFill(0x00FF00);
    bg.graphics.drawRect(0, 0, 100, 200);
    bg.graphics.endFill();

    txt = new TextField();
    txt.border = true;
    txt.text = "Hello World";
}

private function draw():void
{
    addChild(bg);
    addChild(txt);
    txt.x = 50;
    txt.y = 50;
}
}
```

第 31 章：使用外部 API

ActionScript 3.0 外部 API 可实现 ActionScript 与在其中运行 Adobe Flash Player 的容器应用程序之间的直接通信。在某些情况下，您可能需要使用外部 API，例如创建 SWF 文档与 HTML 页中的 JavaScript 之间的交互，或者构建台式机应用程序以使用 Flash Player 来播放 SWF 文件的时候。

本章介绍了如何使用外部 API 与容器应用程序进行交互，如何在 ActionScript 与 HTML 页中的 JavaScript 之间传递数据，以及如何在 ActionScript 和台式机应用程序之间建立通信并交换数据。

注 本章仅讨论 SWF 中的 ActionScript 与容器应用程序之间的通信，该应用程序包含对加载 SWF 的 Flash Player 或 AIR 实例的引用。应用程序中其它任何使用 Flash Player 的情况都不在本文档的讨论范围之内。Flash Player 用作浏览器插件或放映文件（独立应用程序）。可能还在一定程度上支持其它使用方案。

使用外部 API 的基础知识

使用外部 API 简介

虽然在某些情况下可以独立运行 SWF 文件（例如，如果创建 SWF 放映文件），但在大多数情况下，SWF 应用程序作为另一个应用程序中的元素来运行。通常，包含 SWF 的容器是 HTML 文件；在少数情况下，SWF 文件用于台式机应用程序的全部或部分用户界面。

当处理更高级的应用程序时，您可能会发现，需要在 SWF 文件和容器应用程序之间建立通信。例如，网页通常使用 HTML 格式来显示文本或其它信息，并包含 SWF 文件以显示动态可视内容，如图表或视频。在这种情况下，您可能需要建立此类通信，以便当用户单击网页上的按钮时，它将更改 SWF 文件中的某些内容。ActionScript 包含一种称为外部 API 的机制，它有助于在 SWF 文件中的 ActionScript 与容器应用程序中的其它代码之间建立这种类型的通信。

常见外部 API 任务

本章介绍了以下常见的外部 API 任务：

- 获取有关容器应用程序的信息
- 使用 ActionScript 调用容器应用程序中的代码，其中包括网页或台式机应用程序
- 从容器应用程序的代码中调用 ActionScript 代码
- 创建代理以简化从容器应用程序调用 ActionScript 代码的过程

重要概念和术语

以下参考列表包含将会在本章中使用的重要术语：

- **ActiveX 容器：**容器应用程序（不是 Web 浏览器），它包含 Flash Player ActiveX 控件的实例以便在应用程序中显示 SWF 内容。
- **容器应用程序：**Flash Player 在其中运行 SWF 文件的应用程序，如 Web 浏览器和包含 Flash Player 内容的 HTML 页。
- **放映文件：**已转换为独立可执行文件的 SWF 文件，其中包括 Flash Player 以及 SWF 文件的内容。可以使用 Adobe Flash CS4 Professional 或独立的 Flash Player 来创建放映文件。放映文件通常用于以 CD-ROM 形式分发 SWF 文件，或在以下类似情况下进行 SWF 文件分发：下载文件大小不是问题，并且 SWF 作者希望确保用户能够运行 SWF 文件，而无论用户计算机上是否安装了 Flash Player。

- 代理：这是一个中间应用程序或代码，它代表一个应用程序（“调用应用程序”）调用另一个应用程序（“外部应用程序”）中的代码，并将值返回给调用应用程序。可以出于各种不同的原因来使用代理，其中包括：
 - 通过将调用应用程序中的本机函数调用转换为外部应用程序所理解的格式，简化进行外部函数调用的过程
 - 解决禁止调用方直接与外部应用程序进行通信的安全或其它限制问题
- 序列化：将对象或数据值转换为某种格式，这种格式可用于通过消息在两个编程系统之间传递值，如通过 Internet 或在一台计算机上运行的两个不同应用程序之间进行传递。

完成本章中的示例

学习本章的过程中，您可能想要测试示例代码清单。本章中的许多代码清单是用于演示的较小的代码清单，而不是完整的工作示例或用于检查值的完整代码。由于使用外部 API 需要（根据定义）编写 ActionScript 代码以及容器应用程序中的代码，因此测试示例涉及创建一个容器（例如，包含 SWF 的网页）和使用代码清单与该容器交互。

要测试 ActionScript 与 JavaScript 之间的通信的示例，请执行以下操作：

- 1 使用 Flash 创作工具创建一个新文档并将该文档保存到您的计算机上。
- 2 从主菜单中，选择“文件”>“发布设置”。
- 3 在“发布设置”对话框中，在“格式”选项卡上确认选中了“Flash”和“HTML”复选框。
- 4 单击“发布”按钮。这将在同一个文件夹中生成一个 SWF 文件和一个 HTML 文件，且两个文件的名称与您用于保存文档的名称相同。单击“确定”以关闭“发布设置”对话框。
- 5 取消选择“HTML”复选框。现在 HTML 页即生成，您将修改该页来添加适当的 JavaScript 代码。取消选择“HTML”复选框可确保在您修改 HTML 页之后，Flash 在发布 SWF 文件时不会使用新的 HTML 页覆盖您所做的更改。
- 6 单击“确定”以关闭“发布设置”对话框。
- 7 使用 HTML 或文本编辑器应用程序打开 Flash 在您发布 SWF 时创建的 HTML 文件。在 HTML 源代码中，添加开始和结束 script 标签，并将示例代码清单中的 JavaScript 代码复制到其中：

```
<script>
// add the sample JavaScript code here
</script>
```
- 8 保存该 HTML 文件并返回到 Flash。
- 9 选择时间轴的第 1 帧中的关键帧，并打开“动作”面板。
- 10 将 ActionScript 代码清单复制到“脚本”窗格中。
- 11 从主菜单中，选择“文件”>“发布”以使用您所做的更改更新该 SWF 文件。
- 12 使用 Web 浏览器打开您编辑过的 HTML 页，查看该页并测试 ActionScript 与 HTML 页之间的通信。

要测试 ActionScript 与 ActiveX 容器之间的通信的示例，请执行以下操作：

- 1 使用 Flash 创作工具创建一个新文档并将该文档保存到您的计算机上。您可能想要将它保存在容器应用程序预计能够在其中找到 SWF 文件的文件夹中。
- 2 从主菜单中，选择“文件”>“发布设置”。
- 3 在“发布设置”对话框中，在“格式”选项卡上确认仅选中了“Flash”复选框。
- 4 在“Flash”复选框旁边的“文件”字段中，单击文件夹图标以选择 SWF 文件将发布到的文件夹。通过设置 SWF 文件的位置，您可以（举例而言）将创作文档放在一个文件夹中，而将发布的 SWF 文件放在另一个文件夹（例如包含容器应用程序的源代码的文件夹）中。
- 5 选择时间轴的第 1 帧中的关键帧，并打开“动作”面板。
- 6 将示例的 ActionScript 代码复制到“脚本”窗格中。

7 从主菜单中，选择“文件”>“发布”以重新发布该 SWF 文件。

8 创建并运行您的容器应用程序，以测试 ActionScript 与容器应用程序之间的通信。

本章末尾的两个示例分别是一个使用外部 API 与 HTML 页通信的完整示例，和一个使用外部 API 与 C# 台式机应用程序通信的完整示例。这些示例包含完整代码，其中包括 ActionScript 和容器错误检查代码，您在使用外部 API 编写代码时将用到这些代码。有关使用外部 API 的另一个完整示例，请参阅《ActionScript 3.0 语言和组件参考》中 ExternalInterface 类的类示例。

外部 API 要求和优点

外部 API 是 ActionScript 中的一部分，它为在 ActionScript 与充当 Flash Player 容器的“外部应用程序”（通常是 Web 浏览器或独立放映文件应用程序）中所运行的代码之间进行通信提供了一种机制。在 ActionScript 3.0 中，外部 API 的功能是由 ExternalInterface 类提供的。在 Flash Player 8 之前的 Flash Player 版本中，使用 fscommand() 动作与容器应用程序进行通信。ExternalInterface 类替代了 fscommand()，是 JavaScript 与 ActionScript 之间的所有通信的推荐使用机制。

注：如需使用旧的 fscommand() 函数（例如，为了与较早的应用程序保持兼容或与第三方 SWF 容器应用程序或独立的 Flash Player 进行交互），仍可将其作为 flash.system 包中的包级函数来使用。

ExternalInterface 类是一个子系统，通过它可以轻松地实现从 ActionScript 和 Flash Player 到 HTML 页中的 JavaScript 或任何包含 Flash Player 实例的台式机应用程序的通信。

ExternalInterface 类只在以下情况下可用：

- 在所有受支持的 Internet Explorer for Windows 版本（5.0 和更高版本）中
- 在容器应用程序（例如使用 Flash Player ActiveX 控件实例的台式机应用程序）中
- 在支持 NPRuntime 接口的任何浏览器中（当前包括 Firefox 1.0 和更高版本、Mozilla 1.7.5 和更高版本、Netscape 8.0 和更高版本以及 Safari 1.3 和更高版本）。

在其它所有情况下（例如，在独立的播放器中运行），ExternalInterface.available 属性均返回 false。

从 ActionScript 中，可以在 HTML 页上调用 JavaScript 函数。与 fscommand() 相比，外部 API 可提供以下改进功能：

- 可以使用任何 JavaScript 函数，而不仅仅是可与 fscommand() 函数一起使用的函数。
- 可以传递任意数量的、具有任意名称的参数；而不是仅限于传递一个命令和一个字符串参数。这为外部 API 提供了比 fscommand() 大得多的灵活性。
- 可以传递各种数据类型（例如 Boolean、Number 和 String）；不再仅限于 String 参数。
- 可以接收调用值，该值将立即返回到 ActionScript（作为进行的调用的返回值）。

重要说明：如果为 HTML 页中的 Flash Player 实例指定的名称（object 标签的 id 属性）包含有连字符（-）或在 JavaScript 中定义为运算符的其它字符（如 +、*、/、\、.，等等），那么在 Internet Explorer 中查看容器网页的时候将无法从 ActionScript 调用 ExternalInterface。此外，如果定义 Flash Player 实例的 HTML 标签（object 和 embed 标签）嵌套在 HTML form 标签中，也将无法从 ActionScript 调用 ExternalInterface。

使用 ExternalInterface 类

ActionScript 与容器应用程序之间的通信方式有两种：ActionScript 可以调用容器中定义的代码（如 JavaScript 函数），或者容器中的代码可以调用被指定为可调用函数的 ActionScript 函数。在这两种情况下，都可以将信息发送给被调用的代码，而将结果返回给执行调用的代码。

为了便于这种通信，`ExternalInterface` 类包含了两个静态属性和两个静态方法。这些属性和方法可用于获取有关外部接口连接的信息，从 ActionScript 执行容器中的代码，以及使 ActionScript 函数可供容器调用。

获取有关外部容器的信息

`ExternalInterface.available` 属性指示当前的 Flash Player 是否位于提供外部接口的容器中。如果外部接口可用，则此属性为 `true`；否则，为 `false`。在使用 `ExternalInterface` 类中的任何其它功能之前，应始终进行检查以确保当前容器支持外部接口通信，如下所示：

```
if (ExternalInterface.available)
{
    // Perform ExternalInterface method calls here.
}
```

注：`ExternalInterface.available` 属性报告当前容器是否为支持 `ExternalInterface` 连接的容器类型。它不会报告当前浏览器中是否启用了 JavaScript。

通过使用 `ExternalInterface.objectID` 属性（property），您可以确定 Flash Player 实例的唯一标识符（具体来说，是指 Internet Explorer 中 `object` 标签的 `id` 属性（attribute），或者是指使用 `NPRuntime` 接口的浏览器中 `embed` 标签的 `name` 属性（attribute））。这个唯一的 ID 代表浏览器中的当前 SWF 文档，并可用于对 SWF 文档进行引用 — 例如，在容器 HTML 页中调用 JavaScript 函数时进行引用。当 Flash Player 容器不是 Web 浏览器时，此属性为 `null`。

从 ActionScript 中调用外部代码

`ExternalInterface.call()` 方法执行容器应用程序中的代码。它至少需要一个参数，即包含容器应用程序中要调用函数的名称的字符串。传递给 `ExternalInterface.call()` 方法的其它任何参数均作为函数调用的参数传递给容器。

```
// calls the external function "addNumbers"
// passing two parameters, and assigning that function's result
// to the variable "result"
var param1:uint = 3;
var param2:uint = 7;
var result:uint = ExternalInterface.call("addNumbers", param1, param2);
```

如果容器为 HTML 页，此方法将调用具有指定名称的 JavaScript 函数，必须在包含 HTML 页中的 `script` 元素中定义该函数。JavaScript 函数的返回值被传递回 ActionScript。

```
<script language="JavaScript">
    // adds two numbers, and sends the result back to ActionScript
    function addNumbers(num1, num2)
    {
        return (num1 + num2);
    }
</script>
```

如果容器为其它的 ActiveX 容器，此方法将导致 Flash Player ActiveX 控件调度它的 `FlashCall` 事件。Flash Player 将指定的函数名及所有参数序列化为一个 XML 字符串。容器可以在事件对象的 `request` 属性中访问该信息，并用它来确定如何执行它自己的代码。为了将值返回 ActionScript，容器代码会调用 ActiveX 对象的 `SetReturnValue()` 方法，并将结果（序列化为一个 XML 字符串）作为该方法的参数进行传递。有关用于此通信的 XML 格式的详细信息，请参阅第 575 页的“[外部 API 的 XML 格式](#)”。

无论容器为 Web 浏览器还是为其它 ActiveX 容器，只要调用失败或容器方法没有指定返回值，都将返回 `null`。如果包含环境属于调用代码无权访问的安全沙箱，`ExternalInterface.call()` 方法将引发 `SecurityError` 异常。可以通过在包含环境中为 `allowScriptAccess` 设置合适的值来解决此问题。例如，要在 HTML 页中更改 `allowScriptAccess` 的值，请编辑 `object` 和 `embed` 标签中的相应属性。

从容器中调用 ActionScript 代码

容器只能调用函数中的 ActionScript 代码，不能调用任何其它 ActionScript 代码。若要从容器应用程序调用 ActionScript 函数，必须执行两项操作：向 ExternalInterface 类注册该函数，然后从容器的代码调用该函数。

首先，必须注册 ActionScript 函数，指示其应能够为容器所用。使用 ExternalInterface.addCallback() 方法，如下所示：

```
function callMe(name:String) :String
{
    return "busy signal";
}
ExternalInterface.addCallback("myFunction", callMe);
```

addCallback() 方法有两个参数。第一个参数为 String 类型的函数名，容器将籍此名称得知要调用的函数。第二个参数为容器调用定义的函数名时要执行的实际 ActionScript 函数。由于这些名称是截然不同的，因此可以指定将由容器使用的函数名，即使实际的 ActionScript 函数具有不同的名称。这在函数名未知的情况下特别有用 — 例如，指定了匿名函数或需要在运行时确定要调用的函数。

一旦向 ExternalInterface 类注册了 ActionScript 函数，容器就可以实际调用该函数。完成该操作的具体方法依容器的类型而定。例如，在 Web 浏览器的 JavaScript 代码中，使用已注册的函数名调用 ActionScript 函数，就像它是 Flash Player 浏览器对象的方法（即表示 object 或 embed 标签的 JavaScript 对象的方法）。也就是说，将传递参数并返回结果，就如同调用本地函数一样。

```
<script language="JavaScript">
    // callResult gets the value "busy signal"
    var callResult = flashObject.myFunction("my name");
</script>
...
<object id="flashObject" ...>
    ...
    <embed name="flashObject" ... />
</object>
```

或者，在运行于台式机应用程序中的 SWF 文件中调用 ActionScript 函数时，必须将已注册的函数名及所有参数序列化为一个 XML 格式的字符串。然后，将该 XML 字符串作为一个参数来调用 ActiveX 控件的 CallFunction() 方法，以实际执行该调用。有关于此通信的 XML 格式的详细信息，请参阅第 575 页的“[外部 API 的 XML 格式](#)”。

不管是哪种情况，ActionScript 函数的返回值都被传递回容器代码，当调用方为浏览器中的 JavaScript 代码时直接作为值返回，而当调用方为 ActiveX 容器时则会序列化为 XML 格式字符串。

外部 API 的 XML 格式

ActionScript 与承载 Shockwave Flash ActiveX 控件的应用程序间的通信使用特定的 XML 格式对函数调用和值进行编码。外部 API 使用的 XML 格式分为两种。一种格式用于表示函数调用。另一种格式用于表示各个值；此格式用于函数中的参数及函数返回值。函数调用的 XML 格式用于对 ActionScript 的调用和来自 ActionScript 的调用。对于来自 ActionScript 的函数调用，Flash Player 将 XML 传递给容器；而对于来自容器的调用，Flash Player 需要容器应用程序将向其传递一个此格式的 XML 字符串。下面的 XML 片断说明了一个 XML 格式的函数调用示例：

```
<invoke name="functionName" returntype="xml">
    <arguments>
        ...
        (individual argument values)
    </arguments>
</invoke>
```

根节点为 invoke 节点。它具有两个属性：name，指示要调用的函数的名称；以及 returntype，总是为 xml。如果函数调用包括参数，则 invoke 节点具有一个 arguments 子节点，该节点的子节点是使用单个值格式（下面将予以说明）进行了格式设置的参数值。

每个值（包括函数参数和函数返回值）均使用一个格式设置方案，除了实际值之外，该方案还包括数据类型信息。下表列出了 ActionScript 类以及用于对该数据类型的值进行编码的 XML 格式：

ActionScript 类 / 值	C# 类 / 值	格式	注释
null	null	<null/>	
Boolean true	bool true	<true/>	
Boolean false	bool false	<false/>	
String	string	<string> 字符串值 </string>	
Number、int、uint	single、double、int、uint	<number>27.5</number> <number>-12</number>	
Array (元素可以是混合类型)	允许混合类型元素的集合，如 ArrayList 或 object[]	<pre><array> <property id="0"> <number>27.5</number> </property> <property id="1"> <string>Hello there!</string> </property> ... </array></pre>	property 节点定义各个元素，而 id 属性为从零开始的数值索引。
Object	含有字符串键和对象值的字典，如具有字符串键的 HashTable	<pre><object> <property id="name"> <string>John Doe</string> </property> <property id="age"> <string>33</string> </property> ... </object></pre>	property 节点定义各个属性，而 id 属性为属性名称（字符串）。
其它内置或自定义的类		<null/> or <object></object>	ActionScript 将其它对象编码为 null 或空对象。不管是哪种情况，所有属性值都会丢失。

注：该表举例说明了 ActionScript 类，并且还列出了等效 C# 类；但是，外部 API 可用来与支持 ActiveX 控件的任何编程语言或运行时进行通信，而不仅限于 C# 应用程序。

通过将外部 API 用于 ActiveX 容器应用程序来构建您自己的应用程序时，您可能会发现，编写代理以执行将本机函数调用转换为序列化 XML 格式的任务是非常方便的。有关使用 C# 编写的代理类的示例，请参阅第 585 页的“[深入 ExternalInterfaceProxy 类内部](#)”。

示例：将外部 API 用于网页容器

本范例应用程序演示实现 ActionScript 与 Web 浏览器中的 JavaScript 间的通信的正确方法，范例的环境为一个允许用户相互对话的即时消息应用程序（因此该应用程序的名称为：Introvert IM）。它使用外部 API 在网页中的 HTML 表单与 SWF 接口之间发送消息。本示例说明的方法包括：

- 通过验证浏览器在建立通信之前已准备就绪，正确地启动通信
- 检查容器是否支持外部 API
- 从 ActionScript 调用 JavaScript 函数，传递参数，并接收响应中的值
- 使 ActionScript 方法可由 JavaScript 调用，并执行这些调用

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。Introvert IM 应用程序文件位于 Samples/IntrovertIM_HTML 文件夹中。该应用程序包含以下文件：

文件	描述
IntrovertIMApp.fla 或 IntrovertIMApp.mxml	Flash 或 Flex 的主应用程序文件（分别为 FLA 和 MXML）。
com/example/programmingas3/introvertIM/IMManager.as	建立和管理 ActionScript 与容器间的通信的类。
com/example/programmingas3/introvertIM/IMMessageEvent.as	自定义事件类型，从容器接收到消息时由 IMManager 类调度。
com/example/programmingas3/introvertIM/IMStatus.as	枚举，其值表示可在该应用程序中选择的不同“可用性”状态值。
html-flash/IntrovertIMApp.html 或 html-template/index.template.html	Flash 应用程序的 HTML 页 (html-flash/IntrovertIMApp.html)，或用于为 Adobe Flex 应用程序创建容器 HTML 页的模板 (html-template/index.template.html)。此文件包含组成该应用程序的容器部分的所有 JavaScript 函数。

准备 ActionScript 与浏览器间的通信

外部 API 最常见的用途之一就是允许 ActionScript 应用程序与 Web 浏览器进行通信。使用外部 API 时，ActionScript 方法可以调用使用 JavaScript 编写的代码，反之亦然。由于浏览器的复杂性及其内部呈现页的方式，因此根本无法保证 SWF 文档能够在 HTML 页中的第一个 JavaScript 运行之前注册它的回调。出于这个原因，在从 JavaScript 调用 SWF 文档中的函数之前，SWF 文档应总是调用 HTML 页，以通知它 SWF 文档已准备好接受连接。

例如，通过 IMManager 类执行的一系列步骤，Introvert IM 可确定浏览器是否做好了通信的准备，并为通信准备 SWF 文件。第一个步骤（确定浏览器是否已做好通信准备）是在 IMManager 构造函数中执行的，如下所示：

```
public function IMManager(initialStatus:IMStatus)
{
    _status = initialStatus;

    // Check if the container is able to use the external API.
    if (ExternalInterface.available)
    {
        try
        {
            // This calls the isContainerReady() method, which in turn calls
            // the container to see if Flash Player has loaded and the container
            // is ready to receive calls from the SWF.
            var containerReady:Boolean = isContainerReady();
            if (containerReady)
            {
                // If the container is ready, register the SWF's functions.
                setupCallbacks();
            }
            else
            {
                // If the container is not ready, set up a Timer to call the
                // container at 100ms intervals. Once the container responds that
                // it's ready, the timer will be stopped.
                var readyTimer:Timer = new Timer(100);
                readyTimer.addEventListener(TimerEvent.TIMER, timerHandler);
                readyTimer.start();
            }
        }
        ...
    }
    else
    {
        trace("External interface is not available for this container.");
    }
}
```

首先，代码使用 ExternalInterface.available 属性检查外部 API 是否还可在当前容器中使用。如果可用，它将开始建立通信的过程。由于尝试与外部应用程序通信会产生安全异常及其它错误，因此将代码包括在 try 块中（为了简便起见，列表中省略了相应的 catch 块）。

然后，代码调用 isContainerReady() 方法，如下所示：

```
private function isContainerReady():Boolean
{
    var result:Boolean = ExternalInterface.call("isReady");
    return result;
}
```

isContainerReady() 方法又使用 ExternalInterface.call() 方法来调用 JavaScript 函数 isReady()，如下所示：

```

<script language="JavaScript">
<!--
// ----- Private vars -----
var jsReady = false;
...
// ----- functions called by ActionScript -----
// called to check if the page has initialized and JavaScript is available
function isReady()
{
    return jsReady;
}
...
// called by the onload event of the <body> tag
function pageInit()
{
    // Record that JavaScript is ready to go.
    jsReady = true;
}
...
//-->
</script>

```

`isReady()` 函数仅返回 `jsReady` 变量的值。该变量最初为 `false`；一旦触发了网页的 `onload` 事件，该变量的值即更改为 `true`。也就是说，如果 ActionScript 在加载页之前调用 `isReady()` 函数，则 JavaScript 对 `ExternalInterface.call("isReady")` 返回 `false`，因此使得 ActionScript `isContainerReady()` 方法返回 `false`。加载页之后，JavaScript `isReady()` 函数将返回 `true`，从而使得 ActionScript `isContainerReady()` 方法也返回 `true`。

回到 `IMManager` 构造函数中，根据容器的准备情况，将发生两种情况之一。如果 `isContainerReady()` 返回 `true`，则代码调用 `setupCallbacks()` 方法，该方法将完成与 JavaScript 建立通信的过程。另一方面，如果 `isContainerReady()` 返回 `false`，实际上会使该过程陷入停顿状态。创建 `Timer` 对象，并指示其每隔 100 毫秒调用 `timerHandler()` 方法一次，如下所示：

```

private function timerHandler(event:TimerEvent):void
{
    // Check if the container is now ready.
    var isReady:Boolean = isContainerReady();
    if (isReady)
    {
        // If the container has become ready, we don't need to check anymore,
        // so stop the timer.
        Timer(event.target).stop();
        // Set up the ActionScript methods that will be available to be
        // called by the container.
        setupCallbacks();
    }
}

```

每次调用 `timerHandler()` 方法时，它都会再次检查 `isContainerReady()` 方法的结果。初始化容器后，该方法返回 `true`。然后，代码停止 `Timer`，并调用 `setupCallbacks()` 方法来完成与浏览器建立通信的过程。

向 JavaScript 公开 ActionScript 方法

如上例所示，一旦代码确定浏览器已就绪，就将调用 `setupCallbacks()` 方法。此方法准备 ActionScript 以接收来自 JavaScript 的调用，如下所示：

```

private function setupCallbacks():void
{
    // Register the SWF client functions with the container
    ExternalInterface.addCallback("newMessage", newMessage);
    ExternalInterface.addCallback("getStatus", getStatus);
    // Notify the container that the SWF is ready to be called.
    ExternalInterface.call("setSWFIIsReady");
}

```

`setCallBacks()` 方法通过调用 `ExternalInterface.addCallback()` 来注册两个可从 JavaScript 调用的方法，从而完成与容器进行通信的准备任务。在此代码中，第一个参数与 ActionScript 中方法的名称是相同的，该参数就是 JavaScript 所知道的方法的名称（“`newMessage`” 和 “`getStatus`”）。（在本例中，使用不同的名称毫无益处，因此为了简便起见，重复使用相同的名称。）最后，使用 `ExternalInterface.call()` 方法来调用 JavaScript 函数 `setSWFIIsReady()`，该函数通知容器 ActionScript 函数已注册。

从 ActionScript 到浏览器的通信

Introvert IM 应用程序说明一系列在容器页中调用 JavaScript 函数的示例。在最简单的情况下（`setupCallbacks()` 方法的示例），调用 JavaScript 函数 `setSWFIIsReady()` 时不传递任何参数也不接收返回值：

```
ExternalInterface.call("setSWFIIsReady");
```

在 `isContainerReady()` 方法的另一个示例中，ActionScript 调用 `isReady()` 函数，并接受响应中的一个布尔值：

```
var result:Boolean = ExternalInterface.call("isReady");
```

还可以使用外部 API 将参数传递给 JavaScript 函数。以 `IMManager` 类的 `sendMessage()` 方法为例，该方法在用户向他（或她）的“对话伙伴”发送新消息时调用：

```

public function sendMessage(message:String):void
{
    ExternalInterface.call("newMessage", message);
}

```

再次使用 `ExternalInterface.call()` 调用指定的 JavaScript 函数，以通知浏览器有新的消息。另外，消息本身也作为另一个参数传递给 `ExternalInterface.call()`，从而作为参数传递给 JavaScript 函数 `newMessage()`。

从 JavaScript 调用 ActionScript 代码

通信必须是双向的，Introvert IM 应用程序也不例外。不仅 Flash Player IM 客户端调用 JavaScript 来发送消息，而且 HTML 表单也调用 JavaScript 代码来向 SWF 文件发送消息和请求信息。例如，当 SWF 文件通知容器它已建立了联系并做好了通信的准备时，浏览器执行的第一个操作就是调用 `IMManager` 类的 `getStatus()` 方法，以检索 SWF IM 客户端的初始用户可用性状态。此操作是在网页中（在 `updateStatus()` 函数中）执行的，如下所示：

```

<script language="JavaScript">
...
function updateStatus()
{
    if (swfReady)
    {
        var currentStatus = getSWF("IntrovertIMApp").getStatus();
        document.forms["imForm"].status.value = currentStatus;
    }
}
...
</script>

```

代码检查 `swfReady` 变量的值，该变量跟踪 SWF 文件是否已经通知浏览器，它已在 `ExternalInterface` 类中注册了其方法。如果 SWF 文件已准备好接收通信，则下一行代码 (`var currentStatus = ...`) 实际调用 `IMManager` 类中的 `getStatus()` 方法。该行代码中执行三个操作：

- 调用 `getSWF()` JavaScript 函数，返回对表示 SWF 文件的 JavaScript 对象的引用。当 HTML 页中存在多个 SWF 文件时，传递给 `getSWF()` 的参数决定了返回哪一个浏览器对象。传递给该参数的值必须与 `object` 标签的 `id` 属性以及用于包括 SWF 文件的 `embed` 标签的 `name` 属性相匹配。
- 在使用对 SWF 文件的引用时，`getStatus()` 方法就像是 SWF 对象的方法一样被调用。在这种情况下，使用函数名“`getStatus`”，因为它是使用 `ExternalInterface.addCallback()` 注册的 ActionScript 函数名。
- `getStatus()` ActionScript 方法返回一个值，该值被赋给 `currentStatus` 变量，该变量作为 `status` 文本字段的内容（`value` 属性）被赋值。

注：如果您仔细查看这段代码，可能会注意到在 `updateStatus()` 函数的源代码中，调用 `getSWF()` 函数的代码行实际编写如下：
`var currentStatus = getSWF("${application}").getStatus();` \${application} 文本是 HTML 页面模板中的占位符；在 Adobe Flex Builder 3 为应用程序生成实际的 HTML 页面时，此占位符文本会由用作 `object` 标签的 `id` 属性和 `embed` 标签的 `name` 属性的相同文本所替换（在本示例中为 `IntrovertIMApp`）。这是 `getSWF()` 函数所需值。

`sendMessage()` JavaScript 函数说明如何将参数传递给 ActionScript 函数。（`sendMessage()` 是用户按 HTML 页上的“发送”按钮时调用的函数。）

```
<script language="JavaScript">
...
function sendMessage(message)
{
    if (swfReady)
    {
        ...
        getSWF("IntrovertIMApp").newMessage(message);
    }
}
...
</script>
```

`newMessage()` ActionScript 方法需要使用一个参数，因此将 JavaScript `message` 变量作为 JavaScript 代码中 `newMessage()` 方法调用中的参数传递给 ActionScript。

检测浏览器类型

由于各种浏览器在访问内容的方式上存在差异，所以必须总是使用 JavaScript 来检测用户正在运行哪种浏览器，并根据特定于浏览器的语法使用窗口对象或文本对象访问影片，如本例中的 `getSWF()` JavaScript 函数所示：

```
<script language="JavaScript">
...
function getSWF(movieName)
{
    if (navigator.appName.indexOf("Microsoft") != -1)
    {
        return window[movieName];
    }
    else
    {
        return document[movieName];
    }
}
...
</script>
```

如果脚本不检测用户的浏览器类型，则在 HTML 容器中播放 SWF 文件时，用户可能会遇到意外情况。

示例：将外部 API 用于 ActiveX 容器

本示例说明了如何使用外部 API 在 ActionScript 与使用 ActiveX 控件的台式机应用程序之间进行通信。此示例再次使用了 Introvert IM 应用程序（包括 ActionScript 代码乃至相同的 SWF 文件），因此，不再说明外部 API 在 ActionScript 中的用法。熟悉上一个示例将有助于理解本示例。

本示例中的台式机应用程序是通过 Microsoft Visual Studio .NET 使用 C# 语言编写的。本示例的讨论重点是通过 ActiveX 控件使用外部 API 的特定方法。本示例说明如下技术：

- 从承载 Flash Player ActiveX 控件的台式机应用程序调用 ActionScript 函数
- 从 ActionScript 接收函数调用，并在 ActiveX 容器中处理它们
- 使用 proxy 类来隐藏已序列化的 XML 格式的详细信息，Flash Player 使用该格式将消息发送到 ActiveX 容器

若要获取此范例的应用程序文件，请参阅 www.adobe.com/go/learn_programmingAS3samples_flash_cn。Introvert IM C# 文件位于 Samples/IntrovertIM_CSharp 文件夹中。该应用程序包含以下文件：

文件	描述
AppForm.cs	含有 C# Windows Forms 接口的应用程序主文件。
bin/Debug/IntrovertIMApp.swf	该应用程序加载的 SWF 文件。
ExternalInterfaceProxy/ExternalInterfaceProxy.cs	作为 ActiveX 控件的包装以进行 External Interface 通信的类。它为从 ActionScript 进行调用和接收调用提供了机制。
ExternalInterfaceProxy/ExternalInterfaceSerializer.cs	执行将 Flash Player 的 XML 格式消息转换为 .NET 对象的任务的类。
ExternalInterfaceProxy/ExternalInterfaceEventArgs.cs	此文件定义两个 C# 类型（类）：一个自定义委托类和一个事件参数类，ExternalInterfaceProxy 类用它们来向侦听器通知来自 ActionScript 的函数调用。
ExternalInterfaceProxy/ExternalInterfaceCall.cs	此类是一个值对象，表示从 ActionScript 到 ActiveX 容器的函数调用，具有用于函数名和参数的属性。
bin/Debug/IntrovertIMApp.swf	该应用程序加载的 SWF 文件。
obj/AxInterop.ShockwaveFlashObjects.dll, obj/Interop.ShockwaveFlashObjects.dll	Visual Studio .NET 创建的包装程序集，需要使用它们从托管代码访问 Flash Player (Adobe Shockwave® Flash) ActiveX 控件。

Introvert IM C# 应用程序概述

本范例应用程序代表两个相互通信的即时消息客户端程序（一个位于 SWF 文件中，而另一个则是使用 Windows Forms 构建的）。用户界面包含一个 Shockwave Flash ActiveX 控件的实例（在其中加载包含 ActionScript IM 客户端的 SWF 文件）。该接口还包含组成 Windows Forms IM 客户端的若干个文本字段：一个用于输入消息 (MessageText)，一个用于显示在客户端之间发送的消息的文本 (Transcript)，还有一个用于显示 SWF IM 客户端中设置的可用性状态 (Status)。

包含 Shockwave Flash ActiveX 控件

要将 Shockwave Flash ActiveX 控件包含在您自己的 Windows Forms 应用程序中，必须先将其添加到 Microsoft Visual Studio 工具箱中。

将控件添加到工具箱中：

- 打开 Visual Studio 工具箱。
- 右键单击 Visual Studio 2003 中的 Windows Forms 部分或 Visual Studio 2005 中的任何部分。在 Visual Studio 2003 中，从上下文菜单中选择“添加 / 移除项”（Visual Studio 2005 中为“选择项 ...”）。

此操作将打开“自定义工具箱”(2003)/“选择工具箱项”(2005) 对话框。

3 选择“COM 组件”选项卡，该选项卡列出了计算机上所有可用的 COM 组件，包括 Flash Player ActiveX 控件。

4 滚动到 Shockwave Flash Object 并选择它。

若没有列出此项，请确保在系统中安装了 Flash Player ActiveX 控件。

了解 ActionScript 与 ActiveX 容器间的通信

使用外部 API 与 ActiveX 容器应用程序进行的通信类似于与 Web 浏览器之间的通信，但有一个重要的不同之处。如前所述，当 ActionScript 与 Web 浏览器通信时，对于开发人员而言，只需直接调用函数即可；有关如何设置函数调用和响应格式以在播放器和浏览器间进行传递的详细信息都会被隐藏。然而，使用外部 API 与 ActiveX 容器应用程序进行通信时，Flash Player 以特定的 XML 格式向应用程序发送消息（函数调用和返回值），并要求来自容器应用程序的函数调用和返回值使用相同的 XML 格式。ActiveX 容器应用程序的开发人员编写的代码必须能够理解相应格式的函数调用和响应，以及应能够用相应格式创建它们。

Introvert IM C# 示例包含一组类，通过它们可以避免设置消息格式；相反，您可以在调用 ActionScript 函数和接收来自 ActionScript 的函数调用时，使用标准的数据类型。ExternalInterfaceProxy 类与其它辅助类一起提供该功能，并且可以在任何 .NET 项目中重复使用该类以便于进行外部 API 通信。

下面几段代码摘自主应用程序表单 (AppForm.cs)，说明使用 ExternalInterfaceProxy 类实现的更为简化的交互过程：

```
public class AppForm : System.Windows.Forms.Form
{
    ...
    private ExternalInterfaceProxy proxy;
    ...
    public AppForm()
    {
        ...
        // Register this app to receive notification when the proxy receives
        // a call from ActionScript.
        proxy = new ExternalInterfaceProxy(IntrovertIMApp);
        proxy.ExternalInterfaceCall += new ExternalInterfaceCallEventHandler(proxy_ExternalInterfaceCall);
        ...
    }
    ...
}
```

该应用程序声明和创建名为 proxy 的 ExternalInterfaceProxy 实例，传入对用户界面 (IntrovertIMApp) 中的 Shockwave Flash ActiveX 控件的引用。然后，代码注册 proxy.ExternalInterfaceCall() 方法以接收代理的 ExternalInterfaceCall 事件。当函数调用来自 Flash Player 时，ExternalInterfaceProxy 类将调度该事件。C# 代码接收和响应来自 ActionScript 的函数调用的途径便是订阅该事件。

当函数调用来自 ActionScript 时，ExternalInterfaceProxy 实例 (proxy) 接收调用，转换它的 XML 格式，并通知作为代理的 ExternalInterfaceCall 事件的侦听器的对象。对于 AppForm 类，由 proxy.ExternalInterfaceCall() 方法处理该事件，如下所示：

```
/// <summary>
/// Called by the proxy when an ActionScript ExternalInterface call
/// is made by the SWF
/// </summary>
private object proxy_ExternalInterfaceCall(object sender, ExternalInterfaceCallEventArgs e)
{
    switch (e.FunctionCall.FunctionName)
    {
        case "isReady":
            return isReady();
        case "setSWFIsReady":
            setSWFIsReady();
            return null;
        case "newMessage":
            newMessage((string)e.FunctionCall.Arguments[0]);
            return null;
        case "statusChange":
            statusChange();
            return null;
        default:
            return null;
    }
}
...

```

该方法传递一个 `ExternalInterfaceCallEventArgs` 实例（在本例中名为 `e`）。而该对象具有一个 `FunctionCall` 属性，该属性是 `ExternalInterfaceCall` 类的实例。

`ExternalInterfaceCall` 实例是具有两个属性的简单值对象。`FunctionName` 属性包含 ActionScript `ExternalInterface.Call()` 语句中指定的函数名。如果在 ActionScript 中添加了任何参数，则这些参数都包含在 `ExternalInterfaceCall` 对象的 `Arguments` 属性中。在本例中，处理事件的方法只是一个 `switch` 语句，其作用类似于流量管理器。`FunctionName` 属性 (`e.FunctionCall.FunctionName`) 的值决定了将调用 `AppForm` 类的哪个方法。

先前代码清单中的 `switch` 语句的分支说明常见的方法调用情况。例如，任何方法都必须向 ActionScript 返回一个值（例如，`isReady()` 方法调用）或返回 `null`（就像其它方法调用那样）。在 `newMessage()` 方法调用（传递参数 `e.FunctionCall.Arguments[0]`，即 `Arguments` 数组的第一个元素）中，说明了访问从 ActionScript 传递的参数的方法。

使用 `ExternalInterfaceProxy` 类从 C# 调用 ActionScript 函数比从 ActionScript 接收函数调用更为简单明了。若要调用 ActionScript 函数，请使用 `ExternalInterfaceProxy` 实例的 `Call()` 方法，如下所示：

```
/// <summary>
/// Called when the "Send" button is pressed; the value in the
/// MessageText text field is passed in as a parameter.
/// </summary>
/// <param name="message">The message to send.</param>
private void sendMessage(string message)
{
    if (swfReady)
    {
        ...
        // Call the newMessage function in ActionScript.
        proxy.Call("newMessage", message);
    }
}
...
/// <summary>
/// Call the ActionScript function to get the current "availability"
/// status and write it into the text field.
/// </summary>
private void updateStatus()
{
    Status.Text = (string)proxy.Call("getStatus");
}
...
```

如本例所示，ExternalInterfaceProxy 类的 Call() 方法非常类似于 ActionScript 中的对应方法 ExternalInterface.Call()。第一个参数是一个字符串，即要调用的函数的名称。任何其它参数（未在此处显示）也一起传递给 ActionScript 函数。如果 ActionScript 函数返回值，则由 Call() 方法返回该值（如前例所示）。

深入 ExternalInterfaceProxy 类内部

使用 ActiveX 控件的代理包装不一定总是切实可行的，或者您可能想自行编写代理类（例如，使用不同的编程语言或针对不同的平台）。尽管有关创建代理的所有细节不会在此一一介绍，但是了解本例中代理类的内部运作原理却非常有益。

可以使用 Shockwave Flash ActiveX 控件的 CallFunction() 方法，通过外部 API 从 ActiveX 容器中调用 ActionScript 函数。以下从 ExternalInterfaceProxy 类的 Call() 方法摘录的代码显示了此过程：

```
// Call an ActionScript function on the SWF in "_flashControl",
// which is a Shockwave Flash ActiveX control.
string response = _flashControl.CallFunction(request);
```

在摘录的这段代码中，_flashControl 为 Shockwave Flash ActiveX 控件。使用 CallFunction() 方法执行 ActionScript 函数调用。该方法具有一个参数（本例中为 request），该参数是一个包含 XML 格式指令（包括要调用的 ActionScript 函数名称及所有参数）的字符串。从 ActionScript 返回的任何值都被编码为 XML 格式字符串，并作为 CallFunction() 调用的返回值发送回来。在本例中，XML 字符串存储在 response 变量中。

从 ActionScript 接收函数调用是一个多步骤过程。来自 ActionScript 的函数调用会导致 Shockwave Flash ActiveX 控件调度它的 FlashCall 事件，因此要从 SWF 文件接收调用的类（如 ExternalInterfaceProxy 类）需要为该事件定义一个处理函数。在 ExternalInterfaceProxy 类中，该事件处理函数被命名为 _flashControl_FlashCall()，并进行了注册以便在类构造函数中侦听该事件，如下所示：

```
private AxShockwaveFlash _flashControl;

public ExternalInterfaceProxy(AxShockwaveFlash flashControl)
{
    _flashControl = flashControl;
    _flashControl.FlashCall += new _IShockwaveFlashEvents_FlashCallEventHandler(_flashControl_FlashCall);
}

...
private void _flashControl_FlashCall(object sender, _IShockwaveFlashEvents_FlashCallEvent e)
{
    // Use the event object's request property ("e.request")
    // to execute some action.

    ...
    // Return a value to ActionScript;
    // the returned value must first be encoded as an XML-formatted string.
    _flashControl.SetReturnValue(encodedResponse);
}
```

该事件对象 (e) 具有一个 `request` 属性 (`e.request`)，该属性是一个 XML 格式的字符串，它包含有关函数调用的信息，如函数名和参数。容器可以使用该信息来决定要执行哪些代码。在 `ExternalInterfaceProxy` 类中，请求从 XML 格式转换为一个 `ExternalInterfaceCall` 对象，该对象以更易于访问的形式提供相同的信息。`ActiveX` 控件的 `SetReturnValue()` 方法用于将函数结果返回给 `ActionScript` 调用方；必须再次以外部 API 所用的 XML 格式对结果参数进行编码。

`ActionScript` 与承载 `Shockwave Flash ActiveX` 控件的应用程序间的通信使用特定的 XML 格式对函数调用和值进行编码。在 `Introvert IM C#` 示例中，`ExternalInterfaceProxy` 类使应用程序表单中的代码能够直接对发送到 `ActionScript` 或从中接收的值执行运算，并忽略 `Flash Player` 所用的 XML 格式的细节。为了完成该操作，`ExternalInterfaceProxy` 类使用 `ExternalInterfaceSerializer` 类的方法来将 XML 消息实际转换为 `.NET` 对象。`ExternalInterfaceSerializer` 类具有四个公共方法：

- `EncodeInvoke()`: 将函数名和 `C# ArrayList` 类型的参数编码为相应的 XML 格式。
- `EncodeResult()`: 将结果值编码为相应的 XML 格式。
- `DecodeInvoke()`: 对来自 `ActionScript` 的函数调用进行解码。`FlashCall` 事件对象的 `request` 属性被传递给 `DecodeInvoke()` 方法，而且它将调用转换为一个 `ExternalInterfaceCall` 对象。
- `DecodeResult()`: 对作为 `ActionScript` 函数调用的结果接收的 XML 进行解码。

这些方法将 `C#` 值编码为外部 API 的 XML 格式，并将 XML 解码为 `C#` 对象。有关 `Flash Player` 所用 XML 格式的详细信息，请参阅第 575 页的“[外部 API 的 XML 格式](#)”。

第 32 章 : Flash Player 安全性

安全性是 Adobe、用户、网站所有者和内容开发人员关注的焦点。因此，Adobe® Flash® Player 包含了一组安全性规则和控制，以保护用户、网站所有者和内容开发人员的利益。本章讨论了如何在开发应用程序时使用 Flash Player 安全模型。在本章中，除非另有说明，否则假定所述的全部 SWF 文件均使用 ActionScript 3.0 发布，并在 Flash Player 9.0.124.0 或更高版本中运行。

本章旨在对安全性进行一个总的说明，并不全面阐述所有的实现细节、用法方案或使用某些 API 所产生的分歧。有关 Flash Player 安全性概念的详尽介绍，请参阅 Flash Player 开发人员中心主题“安全性”，网址为 www.adobe.com/go/devnet_security_cn。

有关 Adobe® AIR™ 安全问题的信息，请参阅“AIR 安全性”一章，网址为 www.adobe.com/go/learn_air_flash_cn。

Flash Player 安全性概述

Flash Player 安全性大部分基于加载的 SWF 文件、媒体和其它资源的原始域。来自特定 Internet 域（例如 www.example.com）的 SWF 文件始终可以访问该域的所有数据。这些资源放置在相同的安全分组中，该分组称为“安全沙箱”。（有关详细信息，请参阅第 588 页的“[安全沙箱](#)”。）

例如，SWF 文件可以加载 SWF 文件、位图、音频、文本文件以及自身域中的任何其它资源。此外，只要同一域中的两个 SWF 文件都是使用 ActionScript 3.0 编写的，则始终可以在这两个文件之间执行跨脚本访问操作。“跨脚本访问”是指一个 SWF 文件能够使用 ActionScript 访问另一个 SWF 文件中的属性、方法和对象。

对于使用 ActionScript 3.0 编写的 SWF 文件与使用 ActionScript 早期版本编写的 SWF 文件，它们之间不支持跨脚本访问；但是，可以通过使用 LocalConnection 类在这些文件之间进行通信。另外，默认情况下，禁止 SWF 文件跨脚本访问其它域的 ActionScript 3.0 SWF 文件，也禁止其加载其它域的数据；但是可以通过在加载的 SWF 文件中调用 Security.allowDomain() 方法来授予这种访问权。有关详细信息，请参阅第 600 页的“[跨脚本访问](#)”。

默认情况下，以下基本安全性规则始终适用：

- 位于相同安全沙箱中的资源始终可以互相访问。
- 远程沙箱中的 SWF 文件始终不能访问本地文件和数据。

Flash Player 将以下地址视为单个域，并为每个地址设置单独的安全沙箱：

- <http://example.com>
- <http://www.example.com>
- <http://store.example.com>
- <https://www.example.com>
- <http://192.0.34.166>

即使某个指定域（例如 <http://example.com>）映射到特定 IP 地址（例如 <http://192.0.34.166>），Flash Player 也会为它们分别设置单独的安全沙箱。

开发人员可以使用两种基本方法为 SWF 文件授予访问权限，使之能够访问除该 SWF 文件所属沙箱之外的其它沙箱中的资源：

- Security.allowDomain() 方法（请参阅第 595 页的“[作者（开发人员）控制](#)”）
- URL 策略文件（请参阅第 592 页的“[网站控制（策略文件）](#)”）

在 Flash Player 安全模型中，加载内容与提取或访问数据之间存在区别。“内容”被定义为媒体，包括 Flash Player 可以播放的可视化媒体、音频、视频或包含显示媒体的 SWF 文件。“数据”被定义为只有 ActionScript 代码才能访问的内容。内容和数据的加载方式不同。

- 加载内容 — 可以使用 Loader、Sound 和 NetStream 等这样一些类来加载内容。
- 提取数据 — 可以通过使用 Bitmap 对象、BitmapData.draw() 方法、Sound.id3 属性或者 SoundMixer.computeSpectrum() 方法从加载的媒体内容中提取数据。
- 访问数据 — 可以从使用 URLStream、URLLoader、Socket 和 XMLSocket 等类加载的外部文件（如 XML 文件）中直接访问数据。

Flash Player 安全模型针对加载内容和访问数据定义了不同的规则。通常，加载内容的限制要比访问数据的限制少一些。

通常，可以从任意位置加载内容（SWF 文件、位图、MP3 文件和视频），但是如果内容来自执行加载的 SWF 文件所在的域之外的域，则会将内容划分到单独的安全沙箱中。

下面是加载内容的一些限制：

- 默认情况下，本地 SWF 文件（从非网络地址加载的文件，例如用户硬盘上的文件）会被分类到只能与本地文件系统内容交互的沙箱中。这些文件无法从网络加载内容。有关详细信息，请参阅第 589 页的“[本地沙箱](#)”。
- 实时消息传递协议 (RTMP) 服务器可以限制对内容的访问。有关详细信息，请参阅第 600 页的“[使用 RTMP 服务器传送的内容](#)”。

如果加载的媒体为图像、音频或视频，则其安全沙箱之外的 SWF 文件无法访问该媒体的数据（如像素数据和声音数据），除非该 SWF 文件的域已包含在该媒体原始域的 URL 策略文件中。有关详细信息，请参阅第 602 页的“[作为数据访问加载的媒体](#)”。

加载数据的其它格式包括文本文件或 XML 文件，这些文件可使用 URLLoader 对象来加载。同样，这种情况下要访问其它安全沙箱中的任何数据，必须通过原始域中的 URL 策略文件来授予权限。有关详细信息，请参阅第 604 页的“[使用 URLLoader 和 URLStream](#)”。

安全沙箱

客户端计算机可以从很多来源（如外部 Web 站点或本地文件系统）中获取单个 SWF 文件。当 SWF 文件及其它资源（例如共享对象、位图、声音、视频和数据文件）加载到 Flash Player 中时，Flash Player 会根据这些文件和资源的来源单独地将其分配到安全沙箱中。以下各节介绍了 Flash Player 强制执行的规则，这些规则控制着给定沙箱内的 SWF 文件可以访问的内容。

有关 Flash Player 安全性的详细信息，请参阅 Flash Player 开发人员中心主题“安全性”，网址为 www.adobe.com/go/devnet_security_cn。

远程沙箱

Flash Player 将来自 Internet 的资源（包括 SWF 文件）分类到单独的沙箱中，这些沙箱与各自 Web 站点原始域相对应。默认情况下，对这些文件授予访问其自身所在服务器中任何资源的权限。通过显式的 Web 站点许可和作者许可（例如 URL 策略文件和 Security.allowDomain() 方法），可以允许远程 SWF 文件访问其它域的其它数据。有关详细信息，请参阅第 592 页的“[网站控制（策略文件）](#)”和第 595 页的“[作者（开发人员）控制](#)”。

远程 SWF 文件无法加载任何本地文件或资源。

有关 Flash Player 安全性的详细信息，请参阅 Flash Player 开发人员中心主题“安全性”，网址为 www.adobe.com/go/devnet_security_cn。

本地沙箱

“本地文件”是指通过使用 file: 协议或统一命名约定 (UNC) 路径引用的任何文件。本地 SWF 文件放置在四个本地沙箱中的一个内：

- 只能与本地文件系统内容交互的沙箱 — 出于安全性考虑，Flash Player 在默认情况下将所有本地 SWF 文件和资源放置在只能与本地文件系统内容交互的沙箱中。通过此沙箱，SWF 文件可以读取本地文件（例如通过使用 URLLoader 类），但是它们无法以任何方式与网络进行通信。这样可向用户保证本地数据不会泄漏到网络或以其它方式不适当共享。
- 只能与远程内容交互的沙箱 — 编译 SWF 文件时，可以指定该文件作为本地文件运行时拥有网络访问权限（请参阅第 589 页的“[设置本地 SWF 文件的沙箱类型](#)”）。这些文件位于只能与远程内容交互的沙箱中。分配到只能与远程内容交互的沙箱中的 SWF 文件将失去其本地文件访问权限，但允许这些 SWF 文件访问网络中的数据。不过，只有通过 URL 策略文件或调用 Security.allowDomain() 方法来授予操作权限，才允许远程内容交互的 SWF 文件读取源自网络的数据。为授予此类权限，URL 策略文件必须向“所有”域授予权限，方法是使用 <allow-access-from domain="*"/> 或使用 Security.allowDomain("*")。有关详细信息，请参阅第 592 页的“[网站控制（策略文件）](#)”和第 595 页的“[作者（开发人员）控制](#)”。
- 受信任的本地沙箱 — 注册为受信任（由用户或安装程序注册）的本地 SWF 文件放置在受信任的本地沙箱中。系统管理员和用户还可以根据安全注意事项将本地 SWF 文件重新分配（移动）到受信任的本地沙箱，或者从该沙箱中进行重新分配（请参阅第 590 页的“[管理员控制](#)”和第 591 页的“[用户控制](#)”）。分配到受信任的本地沙箱的 SWF 文件可以与其它任何 SWF 文件交互，也可以从任何位置（远程或本地）加载数据。
- AIR 应用程序安全沙箱 — 该沙箱包含在运行 AIR 应用程序时安装的内容。默认情况下，AIR 应用程序沙箱中的文件可以跨脚本访问任何域中的任何文件。但是，AIR 应用程序沙箱以外的文件不会获许跨脚本访问 AIR 文件。默认情况下，AIR 应用程序沙箱中的文件可以加载任何域中的内容和数据。

只能与远程内容交互的沙箱和只能与本地文件系统内容交互的沙箱之间的通信以及只能与本地文件系统内容交互的沙箱和远程沙箱之间的通信是严格禁止的。运行于 Flash Player 中的应用程序或用户 / 管理员不能授予允许此类通信的权限。

在本地 HTML 文件和本地 SWF 文件之间以任一方向访问脚本（例如使用 ExternalInterface 类）均要求涉及的 HTML 文件和 SWF 文件应位于受信任的本地沙箱中。这是因为浏览器的本地安全模型与 Flash Player 本地安全模型不同。

只能与远程内容交互的沙箱中的 SWF 文件无法加载只能与本地文件系统内容交互的沙箱中的 SWF 文件。只能与本地文件系统内容交互的沙箱中的 SWF 文件无法加载只能与远程内容交互的沙箱中的 SWF 文件。

设置本地 SWF 文件的沙箱类型

通过在创作工具中设置文档发布设置，您可以配置只能与本地文件系统内容交互的沙箱或只能与远程内容交互的沙箱的 SWF 文件。

最终用户或计算机管理员可以指定某个本地 SWF 文件是受信任的，以允许该文件从所有域（本地和网络）加载数据。这一点在“全局 Flash Player 信任”目录和“用户 Flash Player 信任”目录中指定。有关详细信息，请参阅第 590 页的“[管理员控制](#)”和第 591 页的“[用户控制](#)”。

有关本地沙箱的详细信息，请参阅第 589 页的“[本地沙箱](#)”。

Security.sandboxType 属性

SWF 文件的作者可以使用只读的静态 Security.sandboxType 属性来确定 Flash Player 向其分配该 SWF 文件的沙箱类型。


Security 类包括表示 Security.sandboxType 属性可能值的常量，如下所示：

- Security.REMOTE — SWF 文件来自 Internet URL，并遵守基于域的沙箱规则。
- Security.LOCAL_WITH_FILE — SWF 文件是本地文件，但尚未受到用户信任，且没有使用网络名称进行发布。此 SWF 文件可以从本地数据源读取数据，但无法与 Internet 进行通信。
- Security.LOCAL_WITH_NETWORK — SWF 文件是本地文件，且尚未受到用户信任，但已使用网络名称进行发布。此 SWF 文件可与 Internet 通信，但不能从本地数据源读取数据。

- Security.LOCAL_TRUSTED — SWF 文件是本地文件，且已使用“设置管理器”或 Flash Player 信任配置文件受到用户信任。此 SWF 文件既可以从本地数据源读取数据，也可以与 Internet 进行通信。
- Security.APPLICATION — SWF 文件在 AIR 应用程序中运行，并且随该应用程序的包（AIR 文件）一起安装。默认情况下，AIR 应用程序沙箱中的文件可以跨脚本访问任何域中的任何文件。但是，AIR 应用程序沙箱以外的文件不会获许跨脚本访问 AIR 文件。默认情况下，AIR 应用程序沙箱中的文件可以加载任何域中的内容和数据。

权限控制

Flash Player 客户端运行时安全模型是围绕 SWF 文件、本地数据和 Internet URL 等这些对象资源设计而成的模型。“资源持有者”是指拥有或使用这些资源的各方。资源持有者可以对其自己的资源进行控制（安全设置），每种资源有四个持有者。Flash Player 对这些控制严格采用一种权利层次结构，如下图所示：



该图说明，如果管理员限制对资源的访问，则任何其他持有者都不能覆盖该限制。

管理员控制

计算机的管理用户（使用管理权限登录的用户）可以应用能影响计算机所有用户的 Flash Player 安全设置。在非企业环境（例如家庭计算机）中，通常只有一个用户，该用户也拥有管理访问权限。即使是在企业环境中，单个用户也可以拥有计算机管理权限。

管理用户控制有两种类型：

- mms.cfg 文件
- “全局 Flash Player 信任”目录

mms.cfg 文件

mms.cfg 文件是一个文本文件，由管理员用于启用或限制用户对各种功能的访问。Flash Player 启动时，将从此文件中读取其安全设置，然后使用这些设置限制功能。mms.cfg 文件包含管理员用于管理隐私控制、本地文件安全性、套接字连接等功能。

SWF 文件可通过调用 Capabilities.avHardwareDisable 和 Capabilities.localFileReadDisable 属性来访问已禁用功能的某些信息。但是，mms.cfg 文件中的大部分设置无法通过 ActionScript 进行查询。

为对计算机强制执行与应用程序无关的安全和隐私策略，只能由系统管理员修改 mms.cfg 文件。mms.cfg 文件不能用于安装应用程序。虽然使用管理权限运行的安装程序可以修改 mms.cfg 文件的内容，但是 Adobe 将此类使用视为违反用户的信任，并且劝告安装程序的创建者决不要修改 mms.cfg 文件。

mms.cfg 文件存储在以下位置：

- Windows: system\Macromed\Flash\mms.cfg
(例如, C:\WINDOWS\system32\Macromed\Flash\mms.cfg)
- Mac: app support/Macromedia/mms.cfg
(例如, /Library/Application Support/Macromedia/mms.cfg)

有关 mms.cfg 文件的详细信息, 请参阅《Flash Player 管理指南》, 网址为 www.adobe.com/go/flash_player_admin_cn。

“全局 Flash Player 信任”目录

管理用户和安装应用程序可以针对所有用户将指定的本地 SWF 文件注册为受信任。这些 SWF 文件会被分配到受信任的本地沙箱。它们可以与任何其它 SWF 文件进行交互, 也可以从任意位置(远程或本地)加载数据。文件在“全局 Flash Player 信任”目录中被指定为受信任, 位置如下:

- Windows: system\Macromed\Flash\FlashPlayerTrust
(例如, C:\WINDOWS\system32\Macromed\Flash\FlashPlayerTrust)
- Mac: app support/Macromedia/FlashPlayerTrust
(例如, /Library/Application Support/Macromedia/FlashPlayerTrust)

“Flash Player 信任”目录可以包含任意数目的文本文件, 每个文件均列出受信任的路径, 一个路径占一行。每个路径可以是单个的 SWF 文件、HTML 文件, 也可以是目录。注释行以 # 号开头。例如, 包含以下文本的 Flash Player 信任配置文件表示将向指定目录及所有子目录中的所有文件授予受信任状态:

```
# Trust files in the following directories:  
C:\Documents and Settings\All Users\Documents\SampleApp
```

信任配置文件中列出的路径应始终是本地路径或 SMB 网络路径。信任配置文件中的任何 HTTP 路径均会被忽略; 只能信任本地文件。

为避免发生冲突, 应为每个信任配置文件指定一个与安装应用程序相应的文件名, 并且使用 .cfg 文件扩展名。

由于开发人员通过安装应用程序分发本地运行的 SWF 文件, 因此可以让安装应用程序向“全局 Flash Player 信任”目录添加一个配置文件, 为要分发的文件授予完全访问权限。安装应用程序必须由拥有管理权限的用户来运行。与 mms.cfg 文件不同, 包含“全局 Flash Player 信任”目录是为了让安装应用程序授予信任权限。管理用户和安装应用程序都可以使用“全局 Flash Player 信任”目录指定受信任的本地应用程序。

此外, 还有适用于单个用户的“Flash Player 信任”目录(请参阅下一节第 591 页的“[用户控制](#)”)。

用户控制

Flash Player 提供三种不同的用户级别权限设置机制:“设置 UI”、“设置管理器”和“用户 Flash Player 信任”目录。

设置 UI 和设置管理器

“设置 UI”是一种用于配置特定域设置的快速交互机制。“设置管理器”显示一个更详细的界面, 并提供全局更改功能, 全局更改可影响对许多域或所有域拥有的权限。另外, 当 SWF 文件请求新的权限, 要求有关安全或隐私的运行时决策时, 程序会显示一些对话框, 用户可以在这些对话框中调整某些 Flash Player 设置。

设置管理器和设置 UI 提供了安全相关选项, 如摄像头和麦克风设置、共享对象存储设置、与旧内容相关的设置等。

注: 在 mms.cfg 文件中所进行的任何设置(请参阅第 590 页的“[管理员控制](#)”)都不会反映在设置管理器中。

有关设置管理器的详细信息, 请访问 www.adobe.com/go/settingsmanager_cn。

“用户 Flash Player 信任”目录

用户和安装应用程序可以将指定的本地 SWF 文件注册为受信任。这些 SWF 文件会被分配到受信任的本地沙箱。它们可以与任何其它 SWF 文件进行交互，也可以从任意位置（远程或本地）加载数据。用户在“用户 Player 信任”目录中将文件指定为受信任，该目录与 Flash 共享对象存储区域的所在目录相同，位置（特定于当前用户）如下：

- Windows: app data\Macromedia\Flash Player\#Security\FlashPlayerTrust

（例如，在 Windows XP 上：C:\Documents and Settings\JohnD\Application Data\Macromedia\Flash Player\#Security\FlashPlayerTrust，在 Windows Vista 上：
C:\Users\JohnD\AppData\Roaming\Macromedia\Flash Player\#Security\FlashPlayerTrust）

在 Windows 中，Application Data 默认为隐藏状态。若要显示隐藏的文件夹和文件，请选择“我的电脑”打开 Windows 资源管理器，选择“工具”>“文件夹选项”，然后选择“查看”选项卡。在“查看”选项卡之下，选择“显示所有文件和文件夹”单选按钮。

- Mac: app data/Macromedia/Flash Player/#Security/FlashPlayerTrust

（例如，/Users/JohnD/Library/Preferences/Macromedia/Flash Player/#Security/FlashPlayerTrust）

这些设置只会影响当前用户，不会影响登录到计算机的其他用户。如果没有管理权限的用户在属于他们自己的系统中安装了某个应用程序，则“用户 Flash Player 信任”目录允许安装程序将该应用程序注册为该用户的受信任程序。

由于开发人员通过安装应用程序分发本地运行的 SWF 文件，因此可以让安装应用程序向“用户 Flash Player 信任”目录添加一个配置文件，为要分发的文件授予完全访问权限。即使在这种情况下，也将“用户 Flash Player 信任”目录文件视为用户控制，原因是用户操作（安装）启动了它。

此外，还有一个“全局 Flash Player 信任”目录，管理用户或安装程序可使用该目录为所有计算机用户注册应用程序（请参阅第 590 页的“[管理员控制](#)”）。

网站控制（策略文件）

要使来自 Web 服务器的数据可用于来自其它域的 SWF 文件，可以在服务器上创建一个策略文件。“策略文件”是位于服务器上特定位置的 XML 文件。

策略文件可影响对许多资源的访问，其中包括以下内容：

- 位图、声音和视频中的数据
- 加载 XML 和文本文件
- 将 SWF 文件从其它安全域导入到执行加载的 SWF 文件所在的安全域
- 对套接字和 XML 套接字连接的访问

ActionScript 对象可实例化两种不同的服务器连接：基于文档的服务器连接和套接字连接。Loader、Sound、URLLoader 和 URLStream 等 ActionScript 对象可实例化基于文档的服务器连接，这些对象均根据 URL 加载文件。ActionScript Socket 和 XMLSocket 对象进行套接字连接，这些对象操作的是数据流而非加载的文档。

由于 Flash Player 支持两种服务器连接，因此有两种策略文件 — URL 策略文件和套接字策略文件。

- 基于文档的连接需要“URL 策略文件”。服务器通过这些文件指示其数据和文档是否可用于特定域或所有域提供的 SWF 文件。
- 套接字连接需要“套接字策略文件”，套接字策略文件可以在较低的 TCP 套接字级别使用 Socket 类和 XMLSocket 类直接启用网络。

Flash Player 要求使用尝试的连接希望使用的协议来传输策略文件。例如，如果将策略文件放置在您的 HTTP 服务器上，则允许其它域中的 SWF 文件从该服务器（作为 HTTP 服务器）加载数据。但是，如果在这台服务器上未提供套接字策略文件，则禁止其它域的 SWF 文件在套接字级别连接到该服务器。换言之，检索策略文件的方法必须与连接方法相匹配。

策略文件的用法和语法将在本节其余部分加以概述，主要讨论为 Flash Player 10 发布的 SWF 文件的用法和语法。（在早期的 Flash Player 版本中，策略文件实现稍有不同，后续的版本加强了 Flash Player 安全性。）有关策略文件的详细信息，请参阅 Flash Player 开发人员中心主题“Flash Player 9 中策略文件的更改”，网址为 www.adobe.com/go/devnet_security_cn。

主策略文件

默认情况下，Flash Player（和 AIR 应用程序沙箱以外的 AIR 内容）首先在服务器的根目录中查找名为 crossdomain.xml 的 URL 策略文件，然后在端口 843 上查找套接字策略文件。这些位置上的文件称为“主策略文件”。（对于套接字连接，Flash Player 还会在与主连接相同的端口上查找套接字策略文件。但是，在该端口上找到的策略文件并不视为主策略文件。）

除指定访问权限外，主策略文件还包含一条“元策略”语句。元策略指定哪些位置可包含策略文件。URL 策略文件的默认元策略为“master-only”，它表示 /crossdomain.xml 是服务器上唯一允许的策略文件。套接字策略文件的默认元策略为“all”，它表示主机上的任何套接字都能提供套接字策略文件。

注：在 Flash Player 9 及更低版本中，URL 策略文件的默认元策略为“all”，它表示任何目录都可以包含策略文件。如果部署的应用程序不是从默认的 /crossdomain.xml 文件而是从其它位置加载策略文件，并且这些应用程序目前可能运行于 Flash Player 10 中，请您（或服务器管理员）务必修改主策略文件以允许其它策略文件。有关如何指定不同元策略的信息，请参阅 Flash Player 开发人员中心主题“Flash Player 9 中策略文件的更改”，网址为 www.adobe.com/go/devnet_security_cn。

SWF 文件可以通过调用 Security.loadPolicyFile() 方法检索其它策略文件名或其它目录位置。但是，如果主策略文件未指定目标位置能提供策略文件，则调用 loadPolicyFile() 无效，即使该位置有策略文件。在尝试需要策略文件的任何网络操作之前，请先调用 loadPolicyFile()。Flash Player 自动将网络请求加入队列并排在对应的策略文件尝试之后。例如，可以在调用 Security.loadPolicyFile() 之后立即进行网络操作。

在检索主策略文件时，Flash Player 会用三秒钟等待服务器响应。如果未接收到响应，Flash Player 则假定主策略文件不存在。但是，对 loadPolicyFile() 的调用没有默认超时值；Flash Player 假定调用的文件存在，在加载文件之前会一直等待。因此，如果要确保加载主策略文件，请使用 loadPolicyFile() 来显式调用主策略文件。

尽管该方法的名称是 Security.loadPolicyFile()，如果不发出需要策略文件的网络调用，也不能加载策略文件。对 loadPolicyFile() 的调用只是告知 Flash Player 到何处查找所需的策略文件。

系统不会通知您何时启动或完成策略文件请求，也没有必要这样做。Flash Player 执行异步策略检查，自动等待连接的启动直到策略文件检索成功完成。

下面几节介绍只适用于 URL 策略文件的相关信息。有关套接字策略文件的详细信息，请参阅第 604 页的“[连接到套接字](#)”。

URL 策略文件范围

URL 策略文件仅适用于从其中加载该文件的目录及其子目录。根目录中的策略文件适用于整个服务器；而从任意子目录加载的策略文件仅适用于该目录及其子目录。

策略文件仅影响对其所在特定服务器的访问。例如，位于 <https://www.adobe.com:8080/crossdomain.xml> 的策略文件仅适用于在端口 8080 通过 HTTPS 对 www.adobe.com 进行的数据加载调用。

在 URL 策略文件中指定访问权限

策略文件包含单个 <cross-domain-policy> 标签，该标签又包含零个或多个 <allow-access-from> 标签。每个 <allow-access-from> 标签包含一个属性 domain，该属性指定一个确切的 IP 地址、一个确切的域或一个通配符域（任何域）。可采用下列两种方式之一来表示通配符域：

- 单个星号 (*)，表示匹配所有域和所有 IP 地址
- 后接后缀的星号，表示只匹配那些以指定后缀结尾的域

后缀必须以点开头。但是，带有后缀的通配符域可以匹配那些只包含后缀但不包含前导点的域。例如，xyz.com 会被看作是 *.xyz.com 的一部分。IP 域规范中不允许使用通配符。

下面的示例显示了一个 URL 策略文件，该文件允许访问源自 *.example.com、www.friendOfExample.com 和 192.0.34.166 的 SWF 文件。

```
<?xml version="1.0"?>
<cross-domain-policy>
    <allow-access-from domain="*.example.com" />
    <allow-access-from domain="www.friendOfExample.com" />
    <allow-access-from domain="192.0.34.166" />
</cross-domain-policy>
```

如果您指定了一个 IP 地址，则将只向使用 IP 语法从该 IP 地址（例如 <http://65.57.83.12/flashmovie.swf>）加载的 SWF 文件授予访问权限，而不向使用域名语法的 SWF 文件授予访问权限。Flash Player 不执行 DNS 解析。

您可以允许访问来自任何域的文档，如下面的示例所示：

```
<?xml version="1.0"?>
<!- http://www.foo.com/crossdomain.xml -->
<cross-domain-policy>
    <allow-access-from domain="*" />
</cross-domain-policy>
```

每个 `<allow-access-from>` 标签还具有可选的 `secure` 属性，其默认值为 `true`。如果您的策略文件在 HTTPS 服务器上，并且要允许非 HTTPS 服务器上的 SWF 文件从 HTTPS 服务器加载数据，则可以将此属性设置为 `false`。

将 `secure` 属性设置为 `false` 可能会危及 HTTPS 提供的安全性。特别是将此属性设置为 `false` 时，会使安全内容受到电子欺骗和窃听攻击。Adobe 强烈建议不要将 `secure` 属性设置为 `false`。

如果要加载的数据在 HTTPS 服务器上，但是加载数据的 SWF 文件在 HTTP 服务器上，Adobe 建议将加载数据的 SWF 文件移至 HTTPS 服务器。这样，可以使安全数据的所有副本得到 HTTPS 的保护。但是，如果决定必须将要执行加载的 SWF 文件保存在 HTTP 服务器上，则需将 `secure="false"` 属性添加到 `<allow-access-from>` 标签，如以下代码所示：

```
<allow-access-from domain="www.example.com" secure="false" />
```

另一种可用于允许访问权限的元素是 `allow-http-request-headers-from` 标签。此元素授权客户端（此客户端承载其它权限域的内容）向您的域发送用户定义的标头。`<allow-access-from>` 标签授权其它域提取您域中的数据，而 `allow-http-request-headers-from` 标签授权其它域将数据以标头的形式发送到您的域中。在下面的示例中，任何域都可以向当前域发送 `SOAPAction` 标头：

```
<cross-domain-policy>
    <allow-http-request-headers-from domain="*" headers="SOAPAction"/>
</cross-domain-policy>
```

如果 `allow-http-request-headers-from` 语句存在于主策略文件中，它将应用于主机上的所有目录。否则，它只应用于包含此语句的策略文件所在的目录和子目录。

预加载策略文件

从服务器加载数据或连接到套接字是一种异步操作。Flash Player 只是等待策略文件完成下载，然后才开始主操作。但是，从图像中提取像素数据或从声音中提取采样数据是一种同步操作。必须先加载策略文件才能提取数据。加载媒体时，请指定媒体检索策略文件：

- 使用 `Loader.load()` 方法时，设置 `context` 参数的 `checkPolicyFile` 属性，该参数是一个 `LoaderContext` 对象。
- 使用 `` 标签在文本字段中嵌入图像时，将 `` 标签的 `checkPolicyFile` 属性设置为 `"true"`，如下所示：

```
<img checkPolicyFile = "true" src = "example.jpg">
```
- 使用 `Sound.load()` 方法时，设置 `context` 参数的 `checkPolicyFile` 属性，该参数是一个 `SoundLoaderContext` 对象。
- 使用 `NetStream` 类时，设置 `NetStream` 对象的 `checkPolicyFile` 属性。

设置上述参数时， Flash Player 首先会检查是否已经为该域下载了任何策略文件。然后，它查找服务器上默认位置中的策略文件，同时检查 `<allow-access-from>` 语句和是否存在元策略。最后，考虑对 `Security.loadPolicyFile()` 方法的所有未处理的调用，以查看它们是否在范围内。

作者（开发人员）控制

用于授予安全权限的主 ActionScript API 是 `Security.allowDomain()` 方法，它将向指定域中的 SWF 文件授予权限。在下面的示例中， SWF 文件向 `www.example.com` 域提供的 SWF 文件授予权限：

```
Security.allowDomain("www.example.com")
```

此方法为下列各项授予权限：

- SWF 文件之间的跨脚本访问（请参阅第 600 页的“[跨脚本访问](#)”）
- 显示列表访问（请参阅第 602 页的“[遍历显示列表](#)”）
- 事件检测（请参阅第 602 页的“[事件安全性](#)”）
- 对 `Stage` 对象的属性和方法的完全访问（请参阅第 601 页的“[Stage 安全性](#)”）

调用 `Security.allowDomain()` 方法的主要目的是为外部域中的 SWF 文件授予权限以访问调用 `Security.allowDomain()` 方法的 SWF 文件的脚本。有关详细信息，请参阅第 600 页的“[跨脚本访问](#)”。

如果将 IP 地址指定为 `Security.allowDomain()` 方法的参数，则不允许任何源自该指定 IP 地址的访问方进行访问。相反，只允许 URL 中包含该指定 IP 地址的访问方进行访问，而不允许其域名映射到该 IP 地址的访问方进行访问。例如，如果域名 `www.example.com` 映射到 IP 地址 `192.0.34.166`，则对 `Security.allowDomain("192.0.34.166")` 的调用不会授予对 `www.example.com` 的访问权限。

可以将通配符 “`*`” 传递给 `Security.allowDomain()` 方法以允许从所有域进行访问。由于这种方式会为“所有”域中的 SWF 文件授予权限，因此请谨慎使用通配符 “`*`”。

ActionScript 还包括一个权限 API，称为 `Security.allowInsecureDomain()`。此方法与 `Security.allowDomain()` 方法的作用相同，只是从安全 HTTPS 连接提供的 SWF 文件调用时，此方法还会允许非安全协议（例如 HTTP）提供的其它 SWF 文件访问执行调用的 SWF 文件。但是，在安全协议（HTTPS）中的文件与非安全协议（例如 HTTP）中的文件之间执行脚本访问操作并不是一种好的安全性做法；这样做会使安全内容受到电子欺骗和窃听攻击。下面是此类攻击的作用方式：由于 `Security.allowInsecureDomain()` 方法允许通过 HTTP 连接提供的 SWF 文件访问安全 HTTPS 数据，因此介人 HTTP 服务器和用户之间的攻击者能够将 HTTP SWF 文件替换为它们自己的文件，这样便可访问您的 HTTPS 数据。

另一种与安全性相关的重要方法是 `Security.loadPolicyFile()` 方法，该方法可让 Flash Player 在非标准位置检查是否存在策略文件。有关详细信息，请参阅第 592 页的“[网站控制（策略文件）](#)”。

限制网络 API

可通过两种方式限制网络 API。为防止恶意行为，需阻止对常用保留端口进行访问；您不能在自己的代码中覆盖这些阻止设置。要控制 SWF 文件访问与其它端口相关的网络功能，可以使用 `allowNetworking` 设置。

阻止的端口

Flash Player 和 Adobe AIR 对某些端口的 HTTP 访问设有限制，与浏览器类似。通常用于非 HTTP 类型服务器的某些标准端口上不允许使用 HTTP 请求。

任何访问网络 URL 的 API 都会受到这些端口阻止限制的约束。只有直接调用套接字的 API 例外，如 `Socket.connect()` 和 `XMLSocket.connect()`，或 `Security.loadPolicyFile()` 调用（这种情况是加载套接字策略文件）。对套接字连接的允许或拒绝是通过在目标服务器上使用套接字策略文件实现的。

下面的列表显示了适用端口阻止的 ActionScript 3.0 API:

FileReference.download()、 FileReference.upload()、 Loader.load()、 Loader.loadBytes()、 navigateToURL()、 NetConnection.call()、 NetConnection.connect()、 NetStream.play()、 Security.loadPolicyFile()、 sendToURL()、 Sound.load()、 URLLoader.load()、 URLStream.load()

端口阻止也适用于共享库的导入、文本字段中 `` 标签的使用、HTML 页中 SWF 文件的加载（使用 `<object>` 和 `<embed>` 标签）。

下面的列表显示了受阻止的端口：

HTTP: 20 (ftp 数据)、 21 (ftp 控制)

HTTP 和 FTP: 1 (tcpmux)、 7 (echo)、 9 (discard)、 11 (sysstat)、 13 (daytime)、 15 (netstat)、 17 (qotd)、 19 (chargen)、 22 (ssh)、 23 (telnet)、 25 (smtp)、 37 (time)、 42 (name)、 43 (nicname)、 53 (domain)、 77 (priv-rjs)、 79 (finger)、 87 (ttylink)、 95 (supdup)、 101 (hostriame)、 102 (iso-tsap)、 103 (gppitnp)、 104 (acr-nema)、 109 (pop2)、 110 (pop3)、 111 (sunrpc)、 113 (auth)、 115 (sftp)、 117 (uucp-path)、 119 (nntp)、 123 (ntp)、 135 (loc-srv / epmap)、 139 (netbios)、 143 (imap2)、 179 (bgp)、 389 (ldap)、 465 (smtp+ssl)、 512 (print / exec)、 513 (login)、 514 (shell)、 515 (printer)、 526 (tempo)、 530 (courier)、 531 (chat)、 532 (netnews)、 540 (uucp)、 556 (remoteefs)、 563 (nntp+ssl)、 587 (smtp)、 601 (syslog)、 636 (ldap+ssl)、 993 (ldap+ssl)、 995 (pop3+ssl)、 2049 (nfs)、 4045 (lockd)、 6000 (x11)

使用 `allowNetworking` 参数

您可以通过在包含 SWF 内容的 HTML 页中的 `<object>` 和 `<embed>` 标签中设置 `allowNetworking` 参数来控制 SWF 文件访问网络的功能。

`allowNetworking` 的可能的值为：

- "all"（默认值）— 在 SWF 中允许所有的网络 API。
- "internal"— SWF 文件可能不调用浏览器导航或浏览器交互 API（在本节后面部分中列出），但是它会调用任何其它网络 API。
- "none"— SWF 文件可能不调用浏览器导航或浏览器交互 API（在本节后面部分中列出），并且它无法使用任何 SWF 到 SWF 通信 API（也在本节后面部分中列出）。

`allowNetworking` 参数主要在 SWF 文件及所在的 HTML 页来自不同的域时使用。当要加载的 SWF 文件与其所在的 HTML 页来自同一个域时，不建议使用 "internal" 或 "none" 值，原因是您不能保证始终同时加载 SWF 文件和想要的 HTML 页。不受信任方可以从您的域中加载未包含在 HTML 中的 SWF 文件，这种情况下，`allowNetworking` 限制不会按预期发挥作用。

调用被禁止的 API 会引发 `SecurityError` 异常。

在包含 SWF 文件引用的 HTML 页的 `<object>` 和 `<embed>` 标签中添加 `allowNetworking` 参数并设置该参数的值，如下面的示例中所示：

```
<object classic="clsid:d27cdb6e-ae6d-11cf-96b8-444553540000"
      Code base="http://fpdownload.macromedia.com/pub/shockwave/cabs/flash/swflash.cab#version=9,0,124,0"
      width="600" height="400" ID="test" align="middle">
<param name="allowNetworking" value="none" />
<param name="movie" value="test.swf" />
<param name="bgcolor" value="#333333" />
<embed src="test.swf" allowNetworking="none" bgcolor="#333333"
       width="600" height="400"
       name="test" align="middle" type="application/x-shockwave-flash"
       pluginspage="http://www.macromedia.com/go/getflashplayer" />
</object>
```

HTML 页也可以使用脚本生成 SWF 嵌入标签。您需要更改脚本以使其插入正确的 allowNetworking 设置。由 Flash 和 Adobe Flex Builder 生成的 HTML 页使用 AC_FL_RunContent() 函数嵌入 SWF 文件的引用。请将 allowNetworking 参数设置添加到该脚本，如下所示：

```
AC_FL_RunContent( ... "allowNetworking", "none", ...)
```

当 allowNetworking 设置为 "internal" 时，下列 API 被阻止：

navigateToURL()、fscommand()、ExternalInterface.call()

当 allowNetworking 设置为 "none" 时，除了上面所列的 API 外，还会禁止下列 API：

sendToURL()、FileReference.download()、FileReference.upload()、Loader.load()、LocalConnection.connect()、
LocalConnection.send()、NetConnection.connect()、NetStream.play()、Security.loadPolicyFile()、SharedObject.getLocal()、
SharedObject.getRemote()、Socket.connect()、Sound.load()、URLLoader.load()、URLStream.load()、XMLSocket.connect()

即使所选的 allowNetworking 设置允许 SWF 文件使用网络 API，可能仍有其它基于安全沙箱限制的限制（请参阅第 588 页的“[安全沙箱](#)”）。

当 allowNetworking 设置为 "none" 时，无法在 TextField 对象 htmlText 属性的 标签中引用外部媒体（这会引发 SecurityError 异常）。

当 allowNetworking 设置为 "none" 时，从导入的共享库添加到 Flash 创作工具（而不是 ActionScript）中的元件在运行时被阻止。

全屏模式安全性

Flash Player 9.0.27.0 和更高版本支持全屏模式，在该模式中，Flash Player 中运行的内容可以填满整个屏幕。要进入全屏模式，需将 Stage 的 displayState 属性设置为 StageDisplayState.FULL_SCREEN 常量。有关详细信息，请参阅第 237 页的“[处理全屏模式](#)”。

对于在浏览器中运行的 SWF 文件，存在一些安全注意事项。

要启用全屏模式，请在包含 SWF 文件引用的 HTML 页的 <object> 和 <embed> 标签中添加 allowFullScreen 参数，并将参数值设置为 "true"（默认值为 "false"），如下例所示：

```
<object classid="clsid:d27cdb6e-ae6d-11cf-96b8-444553540000"
       codebase="http://fpdownload.macromedia.com/pub/shockwave/cabs/flash/swflash.cab#version=9,0,18,0"
       width="600" height="400" id="test" align="middle">
<param name="allowFullScreen" value="true" />
<param name="movie" value="test.swf" />
<param name="bgcolor" value="#333333" />
<embed src="test.swf" allowFullScreen="true" bgcolor="#333333"
       width="600" height="400"
       name="test" align="middle" type="application/x-shockwave-flash"
       pluginspage="http://www.macromedia.com/go/getflashplayer" />
</object>
```

HTML 页也可以使用脚本生成 SWF 嵌入标签。您必须更改该脚本，以便让它能够插入适当的 allowFullScreen 设置。Flash 和 Flex Builder 生成的 HTML 页使用 AC_FL_RunContent() 函数嵌入对 SWF 文件的引用，并且您需要添加 allowFullScreen 参数设置，如下所示：

```
AC_FL_RunContent( ... "allowFullScreen", "true", ...)
```

仅当在响应鼠标事件或键盘事件时才会调用启动全屏模式的 ActionScript。如果在其它情况下调用，Flash Player 会引发异常。

当内容进入全屏模式时，程序会显示一条消息，指导用户如何退出和返回标准模式。该消息将显示几秒钟，然后淡出。

对于在浏览器中运行的内容，在全屏模式下将限制使用键盘。在 Flash Player 9 中，只支持使应用程序返回正常模式（如通过按 Escape 键）的快捷键。用户不能在文本字段中输入文本，也不能在屏幕上导航。在 Flash Player 10 和更高版本中，支持某些非打印键（具体来说是方向键、空格键和 Tab 键）。但仍禁止文本输入。

在独立的播放器或放映文件中始终允许全屏模式。此外，这些环境中还完全支持使用键盘（包括文本输入）。

调用 Stage 对象的 displayState 属性会因任何调用方没有与 Stage 所有者（主 SWF 文件）位于同一安全沙箱而引发异常。有关详细信息，请参阅第 601 页的“[Stage 安全性](#)”。

管理员可以通过在 mms.cfg 文件中设置 FullScreenDisable = 1 对浏览器中运行的 SWF 文件禁用全屏模式。有关详细信息，请参阅第 590 页的“[管理员控制](#)”。

在浏览器中，必须在 HTML 页面中包含 SWF 文件，才能进入全屏模式。

加载内容

SWF 文件可以加载以下内容类型：

- SWF 文件
- 图像
- 声音
- 视频

加载 SWF 文件和图像

使用 Loader 类加载 SWF 文件和图像（JPG、GIF 或 PNG 文件）。除只能与本地文件系统内容交互的沙箱中的 SWF 文件之外，其它所有 SWF 文件都可以从任何网络域加载 SWF 文件和图像。只有本地沙箱中的 SWF 文件才能从本地文件系统中加载 SWF 文件和图像。但是，只能与远程内容交互的沙箱中的文件只能加载位于受信任的本地沙箱或只能与远程内容交互的沙箱中的本地 SWF 文件。只能与远程内容交互的沙箱中的 SWF 文件可加载非 SWF 文件（例如图像）的本地内容，但是无法访问所加载内容中的数据。

从不受信任的来源（如 Loader 对象的根 SWF 文件所在域以外的域）加载 SWF 文件时，您可能需要为 Loader 对象定义遮罩，以防止加载的内容（Loader 对象的子级）绘制到该遮罩之外的 Stage 部分中，如以下代码所示：

```
import flash.display.*;
import flash.net.URLRequest;
var rect:Shape = new Shape();
rect.graphics.beginFill(0xFFFFFFF);
rect.graphics.drawRect(0, 0, 100, 100);
addChild(rect);
var ldr:Loader = new Loader();
ldr.mask = rect;
var url:String = "http://www.unknown.example.com/content.swf";
var urlReq:URLRequest = new URLRequest(url);
ldr.load(urlReq);
addChild(ldr);
```

当调用 Loader 对象的 load() 方法时，可以指定一个 context 参数，该参数是一个 LoaderContext 对象。LoaderContext 类包括三个属性，用于定义如何使用加载的内容的上下文：

- checkPolicyFile：仅当加载图像文件（不是 SWF 文件）时才会使用此属性。如果图像文件所在的域与包含 Loader 对象的文件所在的域不同，则指定此属性。如果将此属性设置为 true，Loader 将检查 URL 策略文件的原始服务器（请参阅第 592 页的“[网站控制（策略文件）](#)”）。如果服务器授予 Loader 域适当权限，则来自 Loader 域中 SWF 文件的 ActionScript 可以访问所加载图像中的数据。换言之，可以使用 Loader.content 属性获取对表示所加载图像的 Bitmap 对象的引用，或使用 BitmapData.draw() 方法访问所加载图像中的像素。

- **securityDomain:** 仅当加载 SWF 文件（不是图像）时才会使用此属性。如果 SWF 文件所在的域与包含 Loader 对象的文件所在的域不同，则指定此属性。对于 **securityDomain** 属性而言，目前仅支持以下两个值：**null**（默认值）和 **SecurityDomain.currentDomain**。如果指定 **SecurityDomain.currentDomain**，则要求加载的 SWF 文件应“导入”到执行加载的 SWF 文件所在的沙箱中，这意味着其运行方式就像它已从执行加载的 SWF 文件自己的服务器中加载一样。只有在位于加载的 SWF 文件的服务器上找到 URL 策略文件时才允许这样做，从而允许执行加载的 SWF 文件所在的域进行访问。如果找到所需的策略文件，则一旦加载开始，加载方和被加载方可自由地互相访问脚本，原因是它们位于同一沙箱中。请注意，多数情况可以通过执行普通加载操作然后让加载的 SWF 文件调用 **Security.allowDomain()** 方法来取代沙箱导入。由于加载的 SWF 文件将位于自己的原始沙箱中，并因而能够访问自己实际服务器上的资源，因此后一种方法会更易于使用。
- **applicationDomain:** 仅当加载使用 ActionScript 3.0 编写的 SWF 文件（不是图像或使用 ActionScript 1.0 或 2.0 编写的 SWF 文件）时才会使用此属性。当加载文件时，可以指定文件应放置在特定的应用程序域中，而不是默认放置在一个新的应用程序域中，这个新的应用程序域是执行加载的 SWF 文件所在应用程序域的子域。请注意，应用程序域是安全域的子单位，因此仅当要加载的 SWF 文件由于以下原因来自您自己的安全域时，才能指定目标应用程序域：该文件来自您自己的服务器，或者使用 **securityDomain** 属性已成功地将该文件导入到您的安全域中。如果指定应用程序域，但加载的 SWF 文件属于其它安全域，则在 **applicationDomain** 中指定的域将被忽略。有关详细信息，请参阅第 547 页的“[使用 Application-Domain 类](#)”。

有关详细信息，请参阅第 258 页的“[指定加载上下文](#)”。

Loader 对象的一个重要属性就是 **contentLoaderInfo** 属性，该属性是一个 **LoaderInfo** 对象。与大部分对象不同，**LoaderInfo** 对象在执行加载的 SWF 文件和被加载的内容之间共享，并且双方始终可以访问该对象。当被加载的内容为 SWF 文件时，它可以通过 **DisplayObject.loaderInfo** 属性访问 **LoaderInfo** 对象。**LoaderInfo** 对象包括诸如加载进度、加载方和被加载方的 URL、加载方和被加载方之间的信任关系等信息及其它信息。有关详细信息，请参阅第 258 页的“[监视加载进度](#)”。

加载声音和视频

除只能与本地文件系统内容交互的沙箱中的那些 SWF 文件之外，所有 SWF 文件都允许从网络来源加载声音和视频，使用 **Sound.load()**、**NetConnection.connect()** 和 **NetStream.play()** 方法即可。

只有本地 SWF 文件才能从本地文件系统加载媒体。只有只能与本地文件系统内容交互的沙箱或受信任的本地沙箱中的 SWF 文件才能访问这些加载文件中的数据。

对加载的媒体还存在一些其它数据访问限制。有关详细信息，请参阅第 602 页的“[作为数据访问加载的媒体](#)”。

使用文本字段中的 **** 标签加载 SWF 文件和图像

通过使用 **** 标签，可以将 SWF 文件和位图加载到文本字段中，如以下代码所示：

```
<img src = 'filename.jpg' id = 'instanceName' >
```

通过使用 **TextField** 实例的 **getImageReference()** 方法，可以访问以这种方式加载的内容，如以下代码所示：

```
var loadedObject:DisplayObject = myTextField.getImageReference('instanceName');
```

但是请注意，以这种方式加载的 SWF 文件和图像会被放入与各自来源相应的沙箱中。

当在文本字段中使用 **** 标签加载图像文件时，通过 URL 策略文件可以允许访问图像中的数据。通过将 **checkPolicyFile** 属性添加到 **** 标签上，可以检查是否存在策略文件，如以下代码所示：

```
<img src = 'filename.jpg' checkPolicyFile = 'true' id = 'instanceName' >
```

当在文本字段中使用 **** 标签加载 SWF 时，可以允许通过调用 **Security.allowDomain()** 方法来访问该 SWF 文件的数据。

当在文本字段中使用 **** 标签加载外部文件时（相对于使用嵌在 SWF 文件中的 **Bitmap** 类），会自动创建一个 **Loader** 对象作为 **TextField** 对象的子对象，并且会将外部文件加载到该 **Loader** 对象中，就如同使用了 ActionScript 中的 **Loader** 对象来加载文件一样。在这种情况下，**getImageReference()** 方法返回自动创建的 **Loader**。由于此 **Loader** 对象与调用代码位于同一安全沙箱中，因此访问此对象不需要任何安全检查。

但是，当引用 `Loader` 对象的 `content` 属性来访问加载的媒体时，需要应用安全性规则。如果内容是图像，则需要实现 URL 策略文件；如果内容是 SWF 文件，则需要让 SWF 文件中的代码调用 `allowDomain()` 方法。

使用 RTMP 服务器传送的内容

Flash Media Server 使用实时媒体协议 (RTMP) 提供数据、音频和视频。SWF 文件通过使用 `NetConnection` 类的 `connect()` 方法并作为参数传递 RTMP URL 来加载此媒体。Flash Media Server 可以根据所请求文件的域来限制连接并防止内容被下载。有关详细信息，请参阅 Flash Media Server 文档。

对于从 RTMP 源加载的媒体，不能使用 `BitmapData.draw()` 和 `SoundMixer.computeSpectrum()` 方法来提取运行时图形和声音数据。


跨脚本访问

如果两个使用 ActionScript 3.0 编写的 SWF 文件来自同一个域，例如，一个 SWF 文件的 URL 是 `http://www.example.com/swfA.swf`，另一个文件的 URL 是 `http://www.example.com/swfB.swf`，则一个 SWF 文件可以检查并修改另一个 SWF 文件中的变量、对象、属性、方法等等，反之亦然。这称为“跨脚本访问”。

在 AVM1 SWF 文件和 AVM2 SWF 文件之间不支持跨脚本访问。AVM1 SWF 文件是使用 ActionScript 1.0 或 ActionScript 2.0 创建的文件。(AVM1 和 AVM2 指的是 ActionScript 虚拟机。)但是，可以使用 `LocalConnection` 类在 AVM1 和 AVM2 之间发送数据。

如果两个使用 ActionScript 3.0 编写的 SWF 文件来自不同的域（例如，`http://siteA.com/swfA.swf` 和 `http://siteB.com/swfB.swf`），则在默认情况下，Flash Player 既不允许 `swfA.swf` 编写 `swfB.swf` 的脚本，也不允许 `swfB.swf` 编写 `swfA.swf` 的脚本。通过调用 `Security.allowDomain()`，一个 SWF 文件可授予其它域中的 SWF 文件编写其脚本的权限。通过调用 `Security.allowDomain("siteA.com")`，`swfB.swf` 向来自 `siteA.com` 的 SWF 文件授予访问其脚本的权限。

在任何跨域的情况下，明确所涉及的双方非常重要。为了便于进行此讨论，我们将执行跨脚本访问的一方称为“访问方”（通常是执行访问的 SWF），将另一方称为“被访问方”（通常是被访问的 SWF）。当 `siteA.swf` 访问 `siteB.swf` 的脚本时，`siteA.swf` 是访问方，`siteB.swf` 是被访问方，如下图所示：



使用 `Security.allowDomain()` 方法建立的跨域权限是不对称的。在上例中，`siteA.swf` 可以访问 `siteB.swf` 的脚本，但 `siteB.swf` 无法访问 `siteA.swf` 的脚本，这是因为 `siteA.swf` 未调用 `Security.allowDomain()` 方法来授予 `siteB.com` 中的 SWF 文件访问其脚本的权限。可以通过使两个 SWF 文件都调用 `Security.allowDomain()` 方法来设置对称权限。

除了防止 SWF 文件受到其它 SWF 文件发起的跨域脚本编写外，Flash Player 还防止 SWF 文件受到 HTML 文件发起的跨域脚本编写。可以通过 `ExternalInterface.addCallback()` 方法建立的回调执行 HTML 到 SWF 的脚本访问。当 HTML 到 SWF 的脚本访问跨域时，被访问的 SWF 文件必须调用 `Security.allowDomain()` 方法（这与访问方是 SWF 文件时一样），否则操作将失败。有关详细信息，请参阅第 595 页的“[作者（开发人员）控制](#)”。

此外，Flash Player 还对 SWF 到 HTML 的脚本访问提供安全控制。有关详细信息，请参阅第 607 页的“[控制外出 URL 访问](#)”。

Stage 安全性

`Stage` 对象的某些属性和方法可用于显示列表中的任何 `sprite` 或影片剪辑。

但是，我们说 `Stage` 对象有一个所有者，即加载的第一个 SWF 文件。默认情况下，`Stage` 对象的以下属性和方法只能用于与舞台所有者位于同一安全沙箱中的 SWF 文件：

属性	方法
<code>align</code>	<code>addChild()</code>
<code>displayState</code>	<code>addChildAt()</code>
<code>frameRate</code>	<code>addEventlistener()</code>
<code>height</code>	<code>dispatchEvent()</code>
<code>mouseChildren</code>	<code>hasEventlistener()</code>
<code>numChildren</code>	<code>setChildIndex()</code>
<code>quality</code>	<code>willTrigger()</code>
<code>scaleMode</code>	
<code>showDefaultContextMenu</code>	
<code>stageFocusRect</code>	
<code>stageHeight</code>	
<code>stageWidth</code>	
<code>tabChildren</code>	
<code>textSnapshot</code>	
<code>width</code>	

为使与 `Stage` 所有者不在同一沙箱中的 SWF 文件能够访问这些属性和方法，舞台所有者 SWF 文件必须调用 `Security.allowDomain()` 方法来允许外部沙箱的域。有关详细信息，请参阅第 595 页的“[作者（开发人员）控制](#)”。

`frameRate` 属性是一种特殊情况：任何 SWF 文件均能读取 `frameRate` 属性。但是，只有位于 `Stage` 所有者的安全沙箱中的文件（或通过调用 `Security.allowDomain()` 方法被授予权限的文件）才能更改该属性。

此外，对于 `Stage` 对象的 `removeChildAt()` 和 `swapChildrenAt()` 方法还存在一些限制，但是这些限制与其它限制不同。要调用这些方法，不需要与 `Stage` 所有者位于同一域中，而是代码必须与受影响的子对象位于同一域中，或者子对象可以调用 `Security.allowDomain()` 方法。

遍历显示列表

一个 SWF 文件能够访问从其它沙箱中加载的显示对象受到一定限制。为使 SWF 文件能够访问由其它沙箱中的另一个 SWF 文件创建的显示对象，被访问的 SWF 文件必须调用 `Security.allowDomain()` 方法向进行访问的 SWF 文件所在的域授予访问权限。有关详细信息，请参阅第 595 页的“[作者（开发人员）控制](#)”。

要访问由 `Loader` 对象加载的 `Bitmap` 对象，图像文件的原始服务器上必须存在 URL 策略文件，并且该 URL 策略文件必须为尝试访问该 `Bitmap` 对象的 SWF 文件所在的域授予访问权限（请参阅第 592 页的“[网站控制（策略文件）](#)”）。

与加载的文件（和 `Loader` 对象）相对应的 `LoaderInfo` 对象包括以下三种属性（定义加载的对象和 `Loader` 对象之间的关系）：`childAllowsParent`、`parentAllowsChild` 和 `sameDomain`。

事件安全性

根据调度事件的显示对象所在的沙箱，与显示列表相关的事件具有一定的安全性访问限制。显示列表中的事件具有冒泡阶段和捕获阶段（在第 207 页的“[处理事件](#)”中介绍）。在冒泡阶段和捕获阶段期间，事件从源显示对象开始迁移，并经过显示列表中的父显示对象。如果父对象与源显示对象位于不同的安全沙箱中，则捕获阶段和冒泡阶段在父对象的下方停止，除非父对象的所有者与源对象的所有者互相信任。这种互相信任可以通过以下方式来建立：

- 1 拥有父对象的 SWF 文件必须调用 `Security.allowDomain()` 方法，以信任拥有源对象的 SWF 文件所在的域。
- 2 拥有源对象的 SWF 文件必须调用 `Security.allowDomain()` 方法，以信任拥有父对象的 SWF 文件所在的域。

与加载的文件（和 `Loader` 对象）相对应的 `LoaderInfo` 对象包括以下两种属性（定义加载的对象和 `Loader` 对象之间的关系）：`childAllowsParent` 和 `parentAllowsChild`。

对于从显示对象以外的对象调度的事件，不存在任何安全检查或与安全相关的含义。

作为数据访问加载的媒体

访问加载的数据可使用诸如 `BitmapData.draw()` 和 `SoundMixer.computeSpectrum()` 等方法。默认情况下，一个安全沙箱中的 SWF 文件无法从另一个沙箱中加载的媒体所呈现或播放的图形或音频对象中获取像素数据或音频数据。但是，可以使用以下方法授予这种权限：

- 在加载的 SWF 文件中，调用 `Security.allowDomain()` 方法可授予对其他域中 SWF 文件的数据访问权限。
- 对于加载的图像、声音或视频，在被加载文件所在的服务器上添加 URL 策略文件。此策略文件必须向试图调用 `BitmapData.draw()` 或 `SoundMixer.computeSpectrum()` 方法的 SWF 文件所在的域授予访问权限，以便从文件提取数据。

以下各节提供有关访问位图、声音和视频数据的详细信息。

访问位图数据

借助 `BitmapData` 对象的 `draw()` 方法，可以将任何显示对象的当前显示像素绘制到 `BitmapData` 对象。这样能够包括 `MovieClip` 对象、`Bitmap` 对象或任何显示对象的像素。要使用 `draw()` 方法将像素绘制到 `BitmapData` 对象，必须满足以下条件：

- 如果是除所加载位图之外的源对象，则该源对象及其（如果是 `Sprite` 或 `MovieClip` 对象）所有子对象必须与调用 `draw()` 方法的对象位于同一域，或者它们必须位于调用方通过调用 `Security.allowDomain()` 方法可访问的 SWF 文件中。
- 如果是已加载的位图源对象，则该源对象必须与调用 `draw()` 方法的对象位于同一域，或者其源服务器必须包含一个授予调用域访问权限的 URL 策略文件。

如果不满足上述条件，则会引发 `SecurityError` 异常。

当使用 Loader 类的 load() 方法加载图像时，可以指定一个 context 参数，该参数是一个 LoaderContext 对象。如果将 LoaderContext 对象的 checkPolicyFile 属性设为 true，则 Flash Player 将在从其中加载图像的服务器上查找 URL 策略文件。如果存在策略文件且该文件允许执行加载的 SWF 文件所在的域进行访问，则会允许该文件访问 Bitmap 对象中的数据，否则就不允许。

此外，还可以通过文本字段中的 标签在加载的图像中指定 checkPolicyFile 属性。有关详细信息，请参阅第 599 页的“[使用文本字段中的 标签加载 SWF 文件和图像](#)”。

访问声音数据

以下与声音相关的 ActionScript 3.0 API 存在一些安全限制：

- SoundMixer.computeSpectrum() 方法 — 对于与声音文件位于同一安全沙箱的 SWF 文件，始终允许使用该方法。对于其它沙箱中的文件，则需经过安全检查。
- SoundMixer.stopAll() 方法 — 对于与声音文件位于同一安全沙箱的 SWF 文件，始终允许使用该方法。对于其它沙箱中的文件，则需经过安全检查。
- Sound 类的 id3 属性 — 对于与声音文件位于同一安全沙箱的 SWF 文件，始终允许使用该属性。对于其它沙箱中的文件，则需经过安全检查。

每个声音都具有两种与之关联的沙箱（一个内容沙箱和一个所有者沙箱）：

- 声音的源域确定内容沙箱，内容沙箱则确定是否可以通过声音的 id3 属性和 SoundMixer.computeSpectrum() 方法提取声音中的数据。
- 启动声音播放的对象确定所有者沙箱，所有者沙箱则确定是否可以使用 SoundMixer.stopAll() 方法停止声音。

当使用 Sound 类的 load() 方法加载声音时，可以指定一个 context 参数，该参数是一个 SoundLoaderContext 对象。如果将 SoundLoaderContext 对象的 checkPolicyFile 属性设为 true，则 Flash Player 将在从其中加载声音的服务器上查找 URL 策略文件。如果存在策略文件且该文件允许执行加载的 SWF 文件所在的域进行访问，则会允许该文件访问 Sound 对象的 id 属性，否则就不允许。此外，设置 checkPolicyFile 属性可以为加载的声音启用 SoundMixer.computeSpectrum() 方法。

可以使用 SoundMixer.areSoundsInaccessible() 方法确定对 SoundMixer.stopAll() 方法的调用是否会因调用方无法访问一个或多个声音所有者的沙箱而停止全部声音。

调用 SoundMixer.stopAll() 方法会停止与 stopAll() 的调用方位于同一所有者沙箱中的那些声音。它还会停止由调用 Security.allowDomain() 方法的 SWF 文件来启动播放的声音，以允许调用 stopAll() 方法的 SWF 文件所在的域进行访问。任何其它声音均不会停止，可以通过调用 SoundMixer.areSoundsInaccessible() 方法来确定此类声音是否存在。

调用 computeSpectrum() 方法要求播放的每个声音应与调用该方法的对象位于同一沙箱中，或者位于已向调用方的沙箱授予权限的源；否则会引发 SecurityError 异常。对于从 SWF 文件库中嵌入声音加载的声音，通过在加载的 SWF 文件中调用 Security.allowDomain() 方法授予权限。对于从非 SWF 文件的源（源自加载的 mp3 文件或视频文件）中加载的声音，源服务器上的 URL 策略文件将授予访问所加载媒体中数据的权限。如果声音是从 RTMP 数据流加载的，则无法使用 computeSpectrum() 方法。

有关详细信息，请参阅第 595 页的“[作者（开发人员）控制](#)”和第 592 页的“[网站控制（策略文件）](#)”

访问视频数据

可以使用 BitmapData.draw() 方法捕获视频当前帧中的像素数据。

视频有两种不同形式：

- RTMP 视频
- 渐进式视频（该视频是从没有 RTMP 服务器的 FLV 文件加载的）

无法使用 BitmapData.draw() 方法访问 RTMP 视频。

当调用 `BitmapData.draw()` 方法，并且 `source` 参数为渐进式视频时，`BitmapData.draw()` 的调用方必须与 FLV 文件位于同一沙箱，或者 FLV 文件所在的服务器上必须存在一个策略文件，用以向执行调用的 SWF 文件所在的域授予访问权限。通过将 `NetStream` 对象的 `checkPolicyFile` 属性设置为 `true`，可以请求下载该策略文件。

加载数据

SWF 文件可以将服务器的数据加载到 ActionScript 中，也可以将 ActionScript 中的数据发送到服务器。加载数据与加载媒体的操作方式不同，原因是加载的信息将直接显示在 ActionScript 中，而不是作为媒体显示。通常，SWF 文件可以从它们自己的域中加载数据。但是，若要从其它域加载数据，它们通常需要策略文件（请参阅第 592 页的“[网站控制（策略文件）](#)”）。

使用 `URLLoader` 和 `URLStream`

可以加载诸如 XML 文件或文本文件等数据。`URLLoader` 和 `URLStream` 类的 `load()` 方法受到 URL 策略文件权限的控制。

如果使用 `load()` 方法从与调用该方法的 SWF 文件所在域不同的域中加载内容，则 Flash Player 会在被加载资源所在的服务器上检查是否存在 URL 策略文件。如果存在策略文件，并且该文件向执行加载的 SWF 文件所在的域授予访问权限，则可以加载数据。

连接到套接字

默认情况下，Flash Player 会从端口 843 查找套接字策略文件。和 URL 策略文件一样，此文件称为“主策略文件”。

在 Flash Player 6 中首次引入策略文件时，并不支持套接字策略文件。与套接字服务器的连接由默认位置中的一个策略文件授权，该默认位置是套接字服务器所在主机的 HTTP 服务器的端口 80。Flash Player 9 仍支持此功能，但 Flash Player 10 不再支持。在 Flash Player 10 中，只有套接字策略文件可授权套接字连接。

与 URL 策略文件类似，套接字策略文件支持元策略语句（用于指定可提供策略文件的端口）。但是，套接字策略文件的默认元策略不是“`master-only`”而是“`all`”。即，除非主策略文件指定更具限制性的设置，否则 Flash Player 假定主机上的任意套接字均可提供套接字策略文件。

默认情况下禁止访问套接字和 XML 套接字连接，即使所要连接的套接字与 SWF 文件位于同一个域中也是如此。可以从下列任意位置提供套接字策略文件，从而允许套接字级别访问：

- 端口 843（主策略文件的位置）
- 与主套接字连接相同的端口
- 主套接字连接端口之外的端口

默认情况下，Flash Player 在端口 843 和主套接字连接所在的端口上查找套接字策略文件。如果要从其它端口提供套接字策略文件，SWF 文件必须调用 `Security.loadPolicyFile()`。

套接字策略文件具有与 URL 策略文件相同的语法，只是前者还必须指定要对哪些端口授予访问权限。如果套接字策略文件来自低于 1024 的端口号，则它可以对任何端口授予访问权限；如果策略文件来自 1024 或更高的端口，则它只能对 1024 端口和更高的端口授予访问权限。允许的端口在 `<allow-access-from>` 标记中的 `to-ports` 属性中指定。单个端口号、端口范围和通配符都是允许值。

下面是套接字策略文件的示例：

```
<?xml version="1.0"?>
<!DOCTYPE cross-domain-policy SYSTEM "http://www.adobe.com/xml/dtds/cross-domain-policy.dtd">
<!-- Policy file for xmlsocket://socks.mysite.com -->
<cross-domain-policy>
    <allow-access-from domain="*" to-ports="507" />
    <allow-access-from domain="*.example.com" to-ports="507,516" />
    <allow-access-from domain="*.example.org" to-ports="516-523" />
    <allow-access-from domain="adobe.com" to-ports="507,516-523" />
    <allow-access-from domain="192.0.34.166" to-ports="*" />
</cross-domain-policy>
```

要从端口 843 或主套接字连接所在的端口上检索套接字策略文件，请调用 `Socket.connect()` 或 `XMLSocket.connect()` 方法。

Flash Player 首先在端口 843 上检查是否有主策略文件。如果找到主策略文件，**Flash Player** 将检查该文件是否含有在目标端口上禁止套接字策略文件的元策略语句。如果未禁止访问，**Flash Player** 首先在主策略文件中查找适当的 `allow-access-from` 语句。如果找不到主策略文件，**Flash Player** 将从主套接字连接所在的端口上查找套接字策略文件。

要从其它位置检索套接字策略文件，请首先调用 `Security.loadPolicyFile()` 方法，调用时使用特殊的 "xmlsocket" 语法，如下所示：

```
Security.loadPolicyFile("xmlsocket://server.com:2525");
```

先调用 `Security.loadPolicyFile()` 方法，然后再调用 `Socket.connect()` 或 `XMLSocket.connect()` 方法。**Flash Player** 随后将一直等待完成策略文件请求，之后再决定是否允许主连接。但是，如果主策略文件指定目标位置不能提供策略文件，则调用 `loadPolicyFile()` 无效，即使该位置有策略文件。

如果要实现套接字服务器，并且需要提供套接字策略文件，则应决定是使用接受主连接的同一端口提供策略文件，还是使用不同的端口来提供策略文件。无论是哪种情况，服务器都必须等到客户端的第一次传输收到之后才能发送响应。

当 **Flash Player** 请求策略文件时，它始终会在建立连接后传输以下字符串：

```
<policy-file-request/>
```

服务器收到此字符串后，即会传输该策略文件。来自 **Flash Player** 的请求始终由 `null` 字节终止，来自服务器的响应也必须由 `null` 字节终止。

程序对于策略文件请求和主连接并不会使用同一连接，因此请在传输策略文件后关闭连接。如果不关闭连接，**Flash Player** 将关闭策略文件连接，之后重新连接以建立主连接。

发送数据

当 SWF 文件中的 ActionScript 代码向服务器或资源发送数据时，将会发生数据发送操作。对于网络域 SWF 文件，始终允许发送数据。本地 SWF 文件则只有在位于受信任的本地沙箱或只能与远程内容交互的沙箱中时，才能向网络地址发送数据。有关详细信息，请参阅第 589 页的 “[本地沙箱](#)”。

可以使用 `flash.net.sendToURL()` 函数向 URL 发送数据。还可以使用其它方法向 URL 发送请求。这些方法包括 `Loader.load()` 和 `Sound.load()` 等加载方法以及 `URLLoader.load()` 和 `URLStream.load()` 等数据加载方法。

上载和下载文件

`FileReference.upload()` 方法可以将用户选择的文件上载到远程服务器。必须先调用 `FileReference.browse()` 或 `FileReferenceList.browse()` 方法，然后再调用 `FileReference.upload()` 方法。

只有在响应鼠标事件或键盘事件时，才能调用启动 `FileReference.browse()` 或 `FileReferenceList.browse()` 方法的 ActionScript。如果在其它情况下调用，**Flash Player** 10 和更高版本会引发异常。

调用 `FileReference.download()` 方法可打开一个对话框，用户可以在该对话框中从远程服务器下载文件。

注：如果服务器要求用户身份验证，则只有在浏览器中运行的 SWF 文件（即使用浏览器插件或 ActiveX 控件的文件）才可以提供对话框来提示用户输入用户名和密码以进行身份验证，并且只适用于下载。**Flash Player** 不允许上载到需要用户身份验证的服务器。

如果执行调用的 SWF 文件位于只能与本地文件系统内容交互的沙箱中，则不允许执行上载和下载操作。

默认情况下，SWF 文件不会在自身所在服务器之外的服务器上执行上载或下载操作。如果其它服务器提供策略文件向执行调用的 SWF 文件所在的域授予访问权限，则执行调用的 SWF 文件可以在其它服务器上执行上载或下载操作。

从导入到安全域的 SWF 文件加载嵌入内容

当加载 SWF 文件时，可以设置用于加载文件的 Loader 对象的 load() 方法中的 context 参数。此参数是一个 LoaderContext 对象。将此 LoaderContext 对象的 securityDomain 属性设置为 Security.currentDomain 时，Flash Player 将在被加载 SWF 文件所在的服务器上检查是否存在 URL 策略文件。如果存在策略文件，并且该文件向执行加载的 SWF 文件所在的域授予访问权限，则可以作为导入媒体加载 SWF 文件。这样，执行加载的文件可以获得对 SWF 文件的库中对象的访问权限。

SWF 文件访问其它安全沙箱中被加载 SWF 文件的类的另一种方法是：使被加载的 SWF 文件调用 Security.allowDomain() 方法，以向执行调用的 SWF 文件所在的域授予访问权限。可以将对 Security.allowDomain() 方法的调用添加到被加载 SWF 文件的主类的构造函数方法中，然后使执行加载的 SWF 文件添加事件侦听器，以便响应由 Loader 对象的 contentLoaderInfo 属性调度的 init 事件。当调度此事件时，被加载的 SWF 文件已经调用构造函数方法中的 Security.allowDomain() 方法，因此被加载 SWF 文件中的类可用于执行加载的 SWF 文件。执行加载的 SWF 文件可以通过调用 Loader.contentLoaderInfo.applicationDomain.getDefinition() 从被加载的 SWF 文件中检索类。

处理旧内容

在 Flash Player 6 中，用于某些 Flash Player 设置的域基于 SWF 文件所在的域的末尾部分。这些设置包括对摄像头和麦克风访问权限、存储配额及永久共享对象存储的设置。

如果 SWF 文件所在的域包含的段数超过两个（如 www.example.com），则会去除该域的第一段（www），并使用该域的剩余部分。因此，在 Flash Player 6 中，www.example.com 和 store.example.com 都使用 example.com 作为这些设置的域。同样，www.example.co.uk 和 store.example.co.uk 都使用 example.co.uk 作为这些设置的域。这样会导致出现问题，使得来自不相关域（如 example1.co.uk 和 example2.co.uk）的 SWF 文件可以访问相同的共享对象。

在 Flash Player 7 及更高版本中，默认情况下是根据 SWF 文件的精确域来选择播放器设置。例如，来自 www.example.com 的 SWF 文件将使用 www.example.com 的播放器设置，来自 store.example.com 的 SWF 文件将使用 store.example.com 的不同播放器设置。

在使用 ActionScript 3.0 编写的 SWF 文件中，当 Security.exactSettings 设置为 true（默认值）时，Flash Player 将针对精确域使用播放器设置。当它设置为 false 时，Flash Player 使用在 Flash Player 6 中使用的域设置。如果将 exactSettings 更改为使用其它值而不是默认值，则必须在要求 Flash Player 选择播放器设置的任何事件（例如，使用摄像头或麦克风，或者检索永久共享对象）发生之前执行此操作。

如果发布了版本 6 的 SWF 文件并通过该版本创建了永久共享对象，则要从使用 ActionScript 3.0 编写的 SWF 中检索这些永久共享对象，必须先将 Security.exactSettings 设置为 false，然后再调用 SharedObject.getLocal()。

设置 LocalConnection 权限

使用 LocalConnection 类可以开发可以互相发送指令的 SWF 文件。LocalConnection 对象只能在运行于同一台客户端计算机上的 SWF 文件之间进行通信，但这些 SWF 文件可以在不同的应用程序中运行。例如，一个 SWF 文件在浏览器中运行，而另一个 SWF 文件在放映文件中运行。

对于每一次 LocalConnection 通信，都存在一个发送方 SWF 文件和一个侦听器 SWF 文件。默认情况下，Flash Player 允许在同一域中的 SWF 文件之间进行 LocalConnection 通信。对于不同沙箱中的 SWF 文件，侦听器必须通过使用 LocalConnection.allowDomain() 方法来允许发送方具有访问权限。作为参数传递到 LocalConnection.allowDomain() 方法的字符串可以包含以下任意项：确切域名、IP 地址和通配符 *。

allowDomain() 方法的格式已更改，与其在 ActionScript 1.0 和 2.0 中的格式不同。在这两个早期版本中，allowDomain() 是可以实现的回调方法。在 ActionScript 3.0 中，allowDomain() 则是调用的 LocalConnection 类的内置方法。由于此更改，allowDomain() 的用法与 Security.allowDomain() 基本相同。

SWF 文件可以使用 LocalConnection 类的 domain 属性确定其所在的域。

控制外出 URL 访问

通过使用以下 ActionScript 3.0 API 可实现外出脚本访问和外出 URL 访问（通过使用 HTTP URL、mailto: 等）：

- flash.system.fscommand() 函数
- ExternalInterface.call() 方法
- flash.net.navigateToURL() 函数

对于本地运行的 SWF 文件，仅当 SWF 文件和包含该文件的网页（如果存在）位于受信任的本地安全沙箱中时，才能成功调用这些方法。如果内容位于只能与远程内容交互的沙箱或只能与本地文件系统内容交互的沙箱中，则对这些方法的调用将失败。

对于不在本地运行的 SWF 文件，所有这些 API 都可与嵌入这些 API 的网页通信，具体取决于下述 AllowScriptAccess 参数的值。flash.net.navigateToURL() 函数还有额外的功能，可以与任何打开的浏览器窗口或框架通信，而不仅仅是与嵌入 SWF 文件的网页通信。有关此功能的详细信息，请参阅第 608 页的“[使用 navigateToURL\(\) 函数](#)”。

HTML 代码中用于加载 SWF 文件的 AllowScriptAccess 参数控制能否从 SWF 文件内执行外出 URL 访问。此参数在 PARAM 或 EMBED 标记内设置。如果不设置 AllowScriptAccess 的值，则仅当 SWF 文件与 HTML 页来自同一个域时才能通信。

AllowScriptAccess 参数可以有 "always"、"sameDomain" 和 "never" 这三个可能值中的一个。

- 当 AllowScriptAccess 为 "always" 时，SWF 文件可以与其嵌入到的 HTML 页进行通信，即使该 SWF 文件来自不同于 HTML 页的域也可以。
- 当 AllowScriptAccess 为 "sameDomain" 时，仅当 SWF 文件与其嵌入到的 HTML 页来自相同的域时，该 SWF 文件才能与该 HTML 页进行通信。此值是 AllowScriptAccess 的默认值。使用此设置，或者不设置 AllowScriptAccess 的值，可以防止一个域中的 SWF 文件访问另一个域的 HTML 页内的脚本。
- 当 AllowScriptAccess 为 "never" 时，SWF 文件将无法与任何 HTML 页进行通信。在 Adobe Flash CS4 Professional 中，不建议使用该值。如果没有在自己的域中提供不受信任的 SWF 文件，则不建议也不应使用该值。如果确实需要使用不受信任的 SWF 文件，则 Adobe 建议您创建一个不同的子域，并将所有不受信任的内容置于其中。

下面的示例说明如何在 HTML 页中设置 AllowScriptAccess 标签以允许对其他域进行外出 URL 访问：

```
<object id='MyMovie.swf' classid='clsid:D27CDB6E-AE6D-11cf-96B8-444553540000'  
codebase='http://download.adobe.com/pub/shockwave/cabs/flash/swflash.cab#version=9,0,0,0' height='100%'  
width='100%'>  
<param name='AllowScriptAccess' value='always' />  
<param name='src' value='MyMovie.swf' />  
<embed name='MyMovie.swf' pluginspage='http://www.adobe.com/go/getflashplayer' src='MyMovie.swf'  
height='100%' width='100%' AllowScriptAccess='never' />  
</object>
```

使用 `navigateToURL()` 函数

除上述 `allowScriptAccess` 参数所指定的安全设置外，`navigateToURL()` 函数还有一个可选的第二参数：`target`。`target` 参数可用于指定将 URL 请求发送到的 HTML 窗口或框架的名称。此类请求还有其它一些安全限制，具体的限制取决于是否将 `navigateToURL()` 用作脚本访问或非脚本访问语句。

对于脚本访问语句，如 `navigateToURL("javascript: alert('Hello from Flash Player.')")`，适用以下规则。

- 如果 SWF 文件是本地受信任的文件，则请求成功。
- 如果目标是 SWF 文件嵌入到的 HTML 页，则适用上述 `allowScriptAccess` 规则。
- 如果目标容纳的内容是加载自 SWF 文件所在的域，则请求成功。
- 如果目标容纳的内容是加载自与 SWF 文件不同的域，并且上述两种条件均不满足，则请求失败。

对于非脚本访问语句（如 HTTP、HTTPS 和 mailto:），如果满足以下所有条件，则请求失败：

- 目标具有下面任一特殊关键字：`"_top"` 或 `"_parent"`。
- SWF 文件位于其它域的网页中。
- 嵌入 SWF 文件时 `allowScriptAccess` 的值不是 `"always"`。

详细信息

有关外出 URL 访问的详细信息，请参阅《ActionScript 3.0 语言和组件参考》中的以下条目：

- `flash.system.fscommand()` 函数
- `ExternalInterface` 类的 `call()` 方法
- `flash.net.navigateToURL()` 函数

共享对象

Flash Player 提供使用“共享对象”的功能，这些对象是永久位于 SWF 文件外部的 ActionScript 对象，它们或者位于用户的本地文件系统中，或者位于远程 RTMP 服务器上。共享对象与 Flash Player 中的其它媒体相似，也划分到安全沙箱中。但是，共享对象的沙箱模型稍有不同，因为共享对象不是可以跨域边界访问的资源，而是始终从共享对象存储区获得的资源，该存储库特定于调用 `SharedObject` 类的方法的每个 SWF 文件的域。通常，共享对象存储区比 SWF 文件所在的域更精确：默认情况下，每个 SWF 文件使用特定于其整个源 URL 的共享对象存储区。

SWF 文件可以使用 `SharedObject.getLocal()` 和 `SharedObject.getRemote()` 方法的 `localPath` 参数，以便使用仅与其部分 URL 关联的共享对象存储区。这样，SWF 文件可以允许与其它 URL 的其它 SWF 文件共享。即使将 `'/'` 作为 `localPath` 参数传递，仍然会指定特定于其自身所在域的共享对象存储区。

用户可通过使用“Flash Player 设置”对话框或“设置管理器”来限制共享对象访问。默认情况下，可以将共享对象创建为每个域最多可以保存 100 KB 的数据。管理用户和用户还可限制写入文件系统的能力。有关详细信息，请参阅第 590 页的“管理员控制”和第 591 页的“用户控制”。

可以通过为 `SharedObject.getLocal()` 方法或 `SharedObject.getRemote()` 方法的 `secure` 参数指定 `true` 来指定共享对象是安全的。请注意有关 `secure` 参数的以下说明：

- 如果此参数设置为 `true`，则 Flash Player 将创建一个新的安全共享对象或获取一个对现有安全共享对象的引用。只可由通过 HTTPS 传递的 SWF 文件对此安全共享对象进行读取或写入，而该 HTTPS 调用 `SharedObject.getLocal()` 并将 `secure` 参数设置为 `true`。
- 如果此参数设置为 `false`，则 Flash Player 将创建一个新的共享对象或获取一个对现有共享对象的引用，可以通过非 HTTPS 连接传递的 SWF 文件对此共享对象或现有共享对象的引用进行读取或写入。

如果执行调用的 SWF 文件不是来自 HTTPS URL，则为 SharedObject.getLocal() 方法或 SharedObject.getRemote() 方法的 secure 参数指定 true 会导致 SecurityError 异常。

对共享对象存储区的选择基于 SWF 文件的源 URL。即使在导入加载和动态加载这两种情况下也是如此，这时 SWF 文件不是源自简单的 URL。导入加载是指在 LoaderContext.securityDomain 属性设置为 SecurityDomain.currentDomain 时加载 SWF 文件。在这种情况下，被加载的 SWF 文件将具有一个伪 URL，该 URL 以执行加载的 SWF 文件所在的域开头，后跟实际的源 URL。动态加载是指使用 Loader.loadBytes() 方法加载 SWF 文件。在这种情况下，被加载的 SWF 文件将具有一个伪 URL，该 URL 以执行加载的 SWF 文件的完整 URL 开头，后跟一个整数 ID。在导入加载和动态加载这两种情况下，都可以使用 LoaderInfo.url 属性检查 SWF 文件的伪 URL。选择共享对象存储区时，可将该伪 URL 完全视为真实的 URL。可以指定使用部分或全部伪 URL 的共享对象 localPath 参数。

用户和管理员可以选择禁止使用“第三方共享对象”。当在 Web 浏览器中执行的任何 SWF 文件的源 URL 与浏览器地址栏中显示的 URL 属于不同的域时，该 SWF 文件可以使用这样的共享对象。用户和管理员可以出于隐私原因选择禁止使用第三方共享对象，从而可以避免跨域跟踪。为避开这种限制，可能希望确保使用共享对象的任何 SWF 文件都只在 HTML 页面结构内部进行加载，这样可以确保 SWF 文件位于浏览器的地址栏中所示的同一个域中。试图使用第三方 SWF 文件中的共享对象时，如果第三方共享对象禁止使用，则 SharedObject.getLocal() 和 SharedObject.getRemote() 方法将返回 null。有关详细信息，请访问 www.adobe.com/products/flashplayer/articles/thirdpartyso。

摄像头、麦克风、剪贴板、鼠标和键盘访问

当 SWF 文件试图使用 Camera.get() 或 Microphone.get() 方法访问用户的摄像头或麦克风时，Flash Player 将显示一个“隐私”对话框，用户可以在该对话框中允许或拒绝对其摄像头或麦克风的访问。用户和管理用户还可以通过 mms.cfg 文件、“设置 UI”和“设置管理器”中的控制基于站点或全局禁用摄像头访问（请参阅第 590 页的“[管理员控制](#)”和第 591 页的“[用户控制](#)”）。用户加以限制后，Camera.get() 和 Microphone.get() 方法均返回 null 值。可以使用 Capabilities.avHardwareDisable 属性确定是已经通过管理方式禁止了 (true) 对摄像头和麦克风的访问，还是允许 (false) 对摄像头和麦克风的访问。

System.setClipboard() 方法允许 SWF 文件用纯文本字符串替换剪贴板内容。这不会带来任何安全性风险。为避免因剪切或复制到剪贴板的密码和其它敏感数据所带来的风险，并未提供相应的 getClipboard() 方法。

运行于 Flash Player 中的应用程序只能监视在其焦点以内发生的键盘和鼠标事件。运行于 Flash Player 中的内容不能检测其它应用程序中的键盘或鼠标事件。

索引

- 符号**
- ^ (尖号) 174
 - __proto__ 31
 - __resolve 31
 - , (逗号) 运算符 40
 - != (不等于) 运算符 119
 - !== (不全等) 运算符 119
 - ? (条件) 运算符 60
 - ? (问号) 174
 - . (点) 元字符 174
 - . (点) 运算符 52, 67
 - . (点) 运算符, XML 191, 197
 - .. (后代存取器) 运算符, XML 197
 - ... (rest) 参数 71
 - "and" 符 (&) 511
 - () (XML 过滤) 运算符 199
 - () (括号) 运算符 53
 - () (小括号) 元字符 174
 -] (右中括号) 174
 - @ (属性标识符) 运算符, XML 191, 199
 - * (通配符) 运算符, XML 199
 - * (星号) 类型注释 42, 44, 48
 - * (星号) 元字符 174
 - / (正斜杠) 173, 174
 - \ (反斜杠)
 - 在正则表达式中 174
 - \ (反斜杠)
 - 在字符串中 117
 - \\" (反斜杠)
 - 在字符串中 117
 - & (和) 511
 - + (加法) 运算符 119
 - + (加号) 元字符 174
 - + (连接) 运算符, XMLList 197
 - += (加法赋值) 运算符 119, 197
 - == 运算符 119
 - ==> 运算符 119
 - >= 运算符 119
 - | (竖线) 178
 - \$ 替换代码 122
 - \$ 元字符 174
- 数字**
- 128 位地址 509
 - 3D 呈现 438
 - 3D 矢量 423
 - 3D 旋转 436
 - 3D 坐标系 424
- A**
- abstract 类 77
 - ActionScript
 - 包括在应用程序中的方法 20
 - 编写工具 21
 - 存储在 ActionScript 文件中 20
 - 构建应用程序 20
 - 关于 31
 - 开发过程 22
 - OOP 支持的历史 96
 - 使用文本编辑器进行编写 22
 - 说明 4
 - 文档 1
 - 新增功能 4
 - 优点 4
 - 与早期版本的兼容性 6
 - ActionScript 1.0 96
 - ActionScript 2.0, 原型链 98
 - ActionScript 虚拟机 (AVM1) 96
 - ActionScript 虚拟机 2 (AVM2) 96, 99
 - ActionScript 中的核心错误类 164
 - addCallback() 方法 601
 - addEventListener() 方法 85, 210, 218
 - addFilterProperty() 方法 351
 - addListener() 方法 210
 - addPropertyArray() 方法 349
 - addTarget() 方法 353
 - allowDomain() 方法
 - 构造函数和 606
 - 关于跨脚本访问 600
 - img 标签和 599
 - 加载上下文 259
 - LocalConnection 类 518
 - 声音和 603
 - allowFullScreen 属性 597
 - allowInsecureDomain() 方法 518
 - allowNetworking 标签 596
 - Alpha 通道遮罩 255
 - AnimatorFactory 类 353
 - application/x-www-form-urlencoded 510
 - ApplicationDomain 类 259, 547, 599
 - apply() 方法 143
 - arguments 对象 69, 70, 71
 - argumentscallee 属性 70
 - arguments.caller 属性 71
 - arguments.length 属性 70
 - Array 类
 - concat() 方法 136
 - 创建实例 131
 - 构造函数 131
 - 构造函数算法 143
 - 关于 130
 - join() 方法 136
 - 扩展 142
 - length 属性 134, 138
 - pop() 方法 133
 - push() 方法 132, 144
 - reverse() 方法 134
 - shift() 方法 133
 - slice() 方法 136
 - sort() 方法 134
 - sortOn() 方法 134, 135
 - splice() 方法 132, 133
 - toString() 方法 136
 - unshift() 方法 132
 - as 运算符 45, 87
 - as3 编译器选项 143
 - AS3 命名空间 100, 143
 - ASCII 字符 116
 - avHardwareDisable 属性 590
 - AVM1 (ActionScript 虚拟机) 96
 - AVM1Movie 类 229
 - AVM2 (ActionScript 虚拟机 2) 96, 99

- 安全性
 请参阅策略文件
 概述 587
allowNetworking 标签 596
 策略文件 592
 导入的 SWF 文件 606
 端口阻止 595
 发送数据 605
 共享对象 606, 608
img 标签 599
 键盘 609
 剪贴板 609
LocalConnection 类 606
 连接到端口 604
 麦克风 606, 609
 全屏模式 597
 RTMP 600
Stage 601
 沙箱 588
 摄像头 606, 609
 设置 UI 和设置管理器 591
 声音 599, 603
 事件相关 602
 视频 599, 603
 鼠标 609
 套接字 604
 图像 602
URLLoader 604
URLStream 604
 位图 602
 文件, 上载和下载 605
 显示列表 602
 作为数据访问加载的媒体 602
 按位逻辑运算符 60
 按位移位运算符 59
- B**
beginGradientFill() 方法 269
 big-endian 字节顺序 519
Bitmap 类 228, 407
BitmapData 对象, 应用滤镜 295
BitmapData 类 407
 八进制数 50
- Boolean** 类
 严格模式下的隐式强制 49
 转换 50
Boolean 数据类型 46
browse() 方法 605
 bubbles 属性 213
ByteArray 类 142
 绑定方法 74, 84
 磅与像素 566
 包
 创建 33
 导入 33
 点语法 52
 点运算符 32, 52
 顶级 32, 33
 关于 32
 嵌套包 33
 保留字 55
 包装对象 43
 被 0 除 47
 背景颜色, 变为不透明 250
 本地存储 522
 笔触 276
 必需的参数 70
 变量
 不允许覆盖 81
 初始化 42, 195
 范围 40
 基本概念 8
 静态 80
 类型 80
 类型注释 40, 43
 默认值 42
 声明 80
 实例 81
 var 语句 39
 未初始化 42
 无类型 32, 42
 编译器选项 101, 143
 编译时类型检查 43
 标识符 35
 标准模式 44, 67
 播放
 监视音频 496
 控制帧速率 236
- 摄像头和 466
 视频 444
 影片剪辑 339
 暂停和恢复播放音频 496
 播客应用程序
 创建 491
 扩展 497
 不等于 (!=) 运算符 119
 捕获阶段 210
 捕获摄像头输入 462
 捕获用户选择的文本 365
 补间动画 347
 不全等 (!==) 运算符 119
 不透明的背景 250
 不支持私有构造函数 82
- C**
call() 方法 (*ExternalInterface* 类) 597, 607
 callee 属性 70
 caller 属性 71
Camera 类 462
 cancelable 属性 212
Capabilities 类 546
Capabilities.avHardwareDisable 属性 590
Capabilities.localFileReadDisable 属性 590
 catch 块 155
charAt() 方法 118
charCodeAt() 方法 118
checkPolicyFile 属性 594
childAllowsParent 属性 602
 class 关键字 76
clearInterval() 函数 113
clearTimeout() 函数 113
Clipboard 类
 generalClipboard 属性 557
setData() 方法 559
setDataHandler() 方法 559
clipboardData 属性 (HTML 复制和粘贴事件) 558
ClipboardFormats 类 558
ClipboardTransferModes 类 559
clone() 方法 (*BitmapData* 类) 410
clone() 方法 (*Event* 类) 214
ColdFusion 515
ColorTransform 类 289

c
 colorTransform 属性 289
 computeSpectrum() 方法 (SoundMixer
类) 600, 602, 603
 concat() 方法
 String 类 119
 connect() 方法
 LocalConnection 类 597
 NetConnection 类 597, 599
 Socket 类 597
 XMLSocket 类 597
 content 属性 (Loader 类) 600
 contentLoaderInfo 属性 258, 606
 contentType 属性 510
 cookie 522
 createBox() 方法 289
 createGradientBox() 方法 269
 CSS
 加载 368
 样式 367
 已定义 360
 currentDomain 属性 606
 currentTarget 属性 213
 参数
 按值或按引用传递 69
 可选或必需 70
 参数, 按引用或值传递 69
 策略文件 592, 604
 checkPolicyFile 属性和 259, 603
 img 标签和 599
 securityDomain 属性和 599
 提取数据 602
 URLLoader 和 URLStream 类 604
 层叠样式表. 请参阅 CSS
 缠绕 276, 278
 场景, 用于区分时间轴 340
 常数 55, 80, 212
 超类 89
 超时限制 565
 乘法运算符 59
 程序, 基本定义 8
 程序流 61
 词汇环境 73
 存储数据 522
 存取器函数, get 和 set 83

错误
 打印 564
 调试工具 154
 ErrorEvent 类 160, 219
 关于处理 150
 基于状态的事件 160
 类型 150, 152
 throw 语句 156
 显示 157
 异步 152
 重新引发 158
 自定义类 159
 错误处理
 策略 154
 常见任务 150
 工具 153
 默认行为 210
 示例 219
 术语 151
 错误类
 ActionScript 164
 ECMAScript 163
 关于 163
 错误事件 160, 219

D
 data 属性 (URLRequest 类) 511
 dataFormat 属性 514
 Date 对象
 创建示例 109
 获取时间值 109
 Date 类
 date 属性 109
 day 属性 109
 fullYear 属性 109
 getMonth() 方法 82, 109
 getMonthUTC() 方法 109
 getTime() 方法 110
 getTimezoneOffset() 方法 110
 构造函数 109
 关于 108
 hours 属性 109
 milliseconds 属性 109
 minutes 属性 109
 month 属性 109

monthUTC 属性 109
 parse() 方法 82
 seconds 属性 109
 setTime() 方法 110
 date 属性 109
 Date() 构造函数 109
 day 属性 109
 decode() 方法 511
 default xml namespace 指令 201
 Delegate 类 216
 delete 运算符 68, 133
 Dictionary 类
 关于 138
 useWeakReference 参数 140
 dispatchEvent() 方法 219
 DisplayObject 类
 blendShader 属性 330
 关于 225, 231
 stage 属性 210
 DisplayObjectContainer 类 225, 229, 232
 displayState 属性 237, 597
 distance() 方法 285
 do..while 循环 65
 DOM 事件规范 207, 209
 domain 属性 (LocalConnection 类) 607
 download() 方法 597, 605
 draw() 方法 259, 598, 600, 602, 603
 drawPath() 277
 drawTriangles() 433
 dynamic 属性 77
 打印
 磅 566
 常见任务 562
 超时 565
 多页, 示例 567
 方向 567
 概念和术语 562
 关于 562
 Rectangle 对象 566
 矢量或位图 565
 缩放 566
 页面 563
 页面高度和宽度 567
 页面属性 565

异常和返回值 564
 指定区域 566
 大于或等于运算符 119
 大于运算符 57, 119
 代码, 包括在应用程序中的方法 20
 淡化显示对象 253
 导出库元件 341
 导入 SWF 文件 606
 等于运算符 60, 119
 递归函数 71
 递减值 58
 第一个加载的 sprite 225, 258
 递增值 58
 点(.) 元字符 174
 点(.) 运算符 52, 67
 点(.) 运算符, XML 191, 197
 点语法 52
 调试 154
 调试器版本, Flash Player 219
 顶点 423
 动画 256
 动态类 46, 67, 79
 动态文本字段 360
 逗号运算符 40
 端口, 在套接字级别连接 604
 对“and”符(&)进行编码 511
 对数组排序 134, 135
 对文本进行格式设置 367, 369
 对象
 基本概念 10
 实例化 15
 对象的字符串表示形式 119
 对象文本 137
 对象引用
 复制和粘贴支持 557
 多个类定义 547
 多态 89

E
 E4X。请参阅 XML
 ECMAScript for XML。请参阅 XML
 ECMAScript 中的核心错误类 163
 Endian.BIG_ENDIAN 519
 Endian.LITTLE_ENDIAN 519
 enterFrame 事件 211

F
 ErrorEvent 类 160, 219
 -es 编译器选项 143
 Event 类
 bubbles 属性 213
 cancelable 属性 212
 clone() 方法 214
 currentTarget 属性 213
 常量 212
 eventPhase 属性 213
 关于 211
 isDefaultPrevented() 方法 214
 method 类别 213
 preventDefault() 方法 210, 214
 stopImmediatePropogation() 方法 214
 stopPropogation() 方法 214
 target 属性 213
 toString() 方法 214
 type 属性 212
 子类 214
 Event.COMPLETE 510
 EventDispatcher 类
 addEventListener() 方法 85, 210
 dispatchEvent() 方法 219
 IEventDispatch 接口和 87
 willTrigger() 方法 219
 引用 52
 eventPhase 属性 213
 exactSettings 属性 (Security 类) 606
 exec() 方法 183
 extends 关键字 89
 ExternalInterface 类 573, 597, 607
 ExternalInterface.addCallback() 方法 601
 二元运算符 56

Flash
 Flash Media Server 600
 Flash Player
 版本 6 96
 调试器版本 219
 IME 和 549
 浏览器中的全屏 237, 446
 与编码的 FLV 的兼容性 442
 在实例之间通信 515
 Flash 时间轴, 添加 ActionScript 20
 Flash 视频。请参阅 FLV
 Flash 文档 2
 flash_proxy 命名空间 37
 flash.display 包
 关于显示编程 224
 过滤和 293
 绘图 API 265
 声音和 473
 位图和 405
 影片剪辑 337
 用户输入和 498
 flash.geom 包 283
 Flex, 何时用于 ActionScript 21
 FLV
 配置以在服务器上托管 467
 文件格式 443
 在 Macintosh 上 468
 focalLength 属性 428
 for each..in 语句 64, 139, 200
 for 循环 63
 for 循环, XML 191, 200
 for..in 语句 63, 139, 200
 frameRate 属性 236
 fromCharCode() 方法 119
 fscommand() 函数 515, 597, 607
 发送事件 207
 fullScreen 事件 238
 fullScreenSourceRect 属性 239
 fullYear 属性 109
 function 对象 76
 function 关键字 66, 81
 Function.apply() 方法 143
 范围
 变量 40
 函数和 68, 73
 Flash 创作, 何时用于 ActionScript 21

- 块级 41
- 全局 73
- 反向运动 354
- 反斜杠 (\) 字符
 - 在正则表达式中 174
 - 反斜杠 (\\\) 字符
 - 在字符串中 117
 - 反转的字符类（在正则表达式中） 177
- 方法
 - 绑定 74, 84
 - 定义 81
 - 覆盖 92
 - getter 和 setter 83, 93
 - 构造函数 81
 - 基本概念 10
 - 静态 82
 - 实例 82
 - 方括号
 - 使用 130
 - 非零规则 279
 - 分层，重新排列 263
 - 分隔符，将字符串拆分为数组 121
 - 分号 53
 - 覆盖 getter 和 setter 93
 - 复合文本 53
 - 服务器，Flash Media 600
 - 服务器端脚本 514
 - 负无穷大 47
 - 复杂值 42
 - 复制和粘贴
 - 延迟呈现 559
 - 传输模式 559
 - 赋值运算符 61
- G**
 - g 标志（在正则表达式中） 180
 - generalClipboard 属性（Clipboard 类） 557
 - GeometricShapes 示例 101
 - getArmatureByName() 方法 356
 - getBoneByName() 方法 356
 - getDefinition() 方法 606
 - getImageReference() 方法 599
 - getLocal() 方法 523, 597, 606, 608
 - getMonth() 方法 82, 109
 - getMonthUTC() 方法 109
- getRect() 方法 288
- getRemote() 方法 523, 597, 608
- getter 和 setter
 - 覆盖 93
 - 关于 83
- getTime() 方法 110
- getTimer() 函数 113
- getTimezoneOffset() 方法 110
- GIF 图形 257
- Graphics 类
 - beginShaderFill() 方法 327
- GraphicsPathCommand 类 277
- GraphicsStroke 276
- 公共类 34
- 共享对象
 - 安全性和 524, 608
- Flash Player 设置和 606
 - 关于 522
 - 显示内容 524
- 构造函数
 - ActionScript 1.0 中 96
 - 关于 81
 - 固定属性继承 99
 - 骨骼 354
 - 骨架 354
 - 关键字 54
 - 关系运算符 59
 - 光标，自定义 502
 - 滚动文本 363, 364
 - 过滤 XML 数据 199
- H**
 - hours 属性 109
 - HTML 文本
 - 和 CSS 367
 - 显示 362
 - HTMLLoader 类
 - 复制和粘贴 557
 - htmlText 属性 362
 - HTTP 隧道 520
 - 哈希 137, 138
 - 函数
 - arguments 对象 69
 - 参数 69
 - 存取器 83
- 递归 71
- 调用 66
- 对象 72
- 返回值 68
- 范围 68, 73
- 关于 65
- 计时 112
- 将属性添加到 73
- 匿名 66, 71
- 嵌套 69, 73
- 小括号 66
- 函数闭包 65, 69, 73
- 函数表达式 66
- 函数参数 69
- 函数语句 66
- 横向打印 567
- 后代存取器（..）运算符，XML 197
- 后缀运算符 58
- 缓存滤镜和位图 296
- 换行符 117
- 换页符 117
- 回调方法
 - 处理 451
 - 视频提示点和元数据 450
- J**
 - i 标志（在正则表达式中） 180
- Java 套接字服务器 520
- 基本概念
 - 变量 8
 - 创建对象实例 15
- 对象 10
 - 方法 10
 - 流控制 17
 - 事件 11
 - 示例 18
- 属性 10
 - 运算符 16
 - 注释 17
- 继承
 - 固定属性 99
 - 静态属性 94
- 实例属性 90
 - id3 属性 603
- IDataInput 和 IDataOutput 接口 519

IEventDispatcher 接口 87, 217, 218
if 语句 61
if..else 语句 61
IGraphicsData 接口 280
 几何结构
 常见使用任务 283
 概念和术语 265, 284
 关于 283
 激活对象 73
IK 354
IKEEvent 类 358
IKMover 类 357
 基类 89
IME
 合成事件 552
 检查可用性 549
 在 Flash Player 中运用 549
 IME 转换模式
 确定 550
 设置 550
import 语句 34
indexOf() 方法 121
 indices 433
init 事件 211
initFilters() 方法 351
instanceof 运算符 45
int 类, 转换 49
int 数据类型 47
InteractiveObject 类 229
internal 属性 34, 35, 79
intersection() 方法 288
intersects() 方法 288
join() 方法 136
 JPG 图形 257
 IPv6 509
is 运算符 45, 87
isDefaultPrevented() 方法 214
isNaN() 全局函数 42
 计时函数 112
 计时器 110
 基于状态的错误事件 160
 基元类型, 隐式转换 49
 基元值 31, 42
 加法 (+) 运算符 119
 加法赋值 (+=) 运算符 119
 加法运算符 59
 加号 (+) 174
 加载的媒体, 作为数据访问 602
 加载的字节数 258
 加载对象的 URL 258
 加载进度 258
 加载上下文 258
 加载图形 257
 加载文件数据 526
 键, 字符串 137
 渐变 269
 尖号 (^) 字符 174
 键控代码 500
 键盘安全性 609
 键盘输入, 捕获 499
 兼容性, Flash Player 和 FLV 文件 442
 剪贴板
 安全性 609
 保存文本 546
 复制和粘贴 557
 数据格式 558, 559
 系统 557
 将数据保存到文件 527
 脚本超时限制 565
 焦点, 在交互中管理 503
 焦距 435
 结合律, 规则 56
 接口
 定义 87, 89
 关于 87
 扩展 88
 在类中实现 88
 结束语句 53
 静态变量 80
 静态方法 82
 静态属性
 继承 94
 声明 78
 XML 192
 作用域链中 95
 静态文本
 创建 229
 访问 372
 静态文本字段 360
 局部变量 40
 句点 (.). 请参阅点
 句法关键字 54
K
 开发
 过程 22
 计划 20
 客户端系统环境
 常见任务 544
 关于 544
 可视对象。请参阅显示对象
 可选的参数 70
 空白 192
 库元件, 导出 341
 跨脚本访问 600
 跨域策略文件
 请参阅策略文件
 块级作用域 41
 快捷菜单 (上下文菜单) 503
 括号
 XML 过滤运算符 199
L
lastIndexOf() 方法 121
length 属性
 arguments 对象 70
Array 类 134
 字符串 118
level 属性 219
 垃圾回收 67, 140
lineGradientStyle() 方法 269
 little-endian 字节顺序 519
load() 方法 (Loader 类) 258, 594, 597
load() 方法 (Sound 类) 594, 597, 599, 605
load() 方法 (URLLoader 类) 510, 597
load() 方法 (URLStream 类) 597, 605
loadBytes() 方法 258, 594
Loader 类 257, 597, 603, 606
Loader.load() 596
Loader.loadBytes() 596
LoaderContext 对象 594
LoaderContext 类 258, 598, 603
LoaderInfo 类
 监视加载进度 258
 显示对象访问权限 602

loaderInfo 属性 258
loadPolicyFile() 方法 597
LocalConnection 类
 connectionName 参数 519
 关于 515
 权限 606
 受限制 597
LocalConnection.allowDomain() 方法 518, 607
LocalConnection.allowInsecureDomain() 方法 518
LocalConnection.client 属性 516
LocalConnection.connect() 方法 597
localFileReadDisable 属性 590
localToGlobal() 方法 285
 类
 不支持 abstract 77
 创建自定义 23
dynamic 属性 77
 顶级语句 77
 定义 76
 定义内部命名空间 77
 动态 46, 67, 79
 公共类 34
 关于 76
 关于编写代码 23
 继承实例属性 90
 基类 89
internal 属性 79
 静态属性 94
 类体 77
 封闭 46
 默认访问控制 80
 内置 31
private 属性 78
protected 属性 79
public 属性 78
 声明静态属性和实例属性 78
 属性 77
 属性 (property) 属性 (attribute) 78
 私有类 32
 特性 10
 子类 89
 类定义, 多个 547
 类对象 31, 98
 类继承 99
 类路径 34
 类型。请参阅数据类型
 类型不匹配 43
 类型检查
 编译时 43
 运行时 44
 类型注释 40, 43
 类型转换 48, 49, 202
 连接 354
 XML 对象 197
 字符串 119
 连接 (+) 运算符, XMLList 197
 列表外的显示对象 230
 流控制, 基本概念 17
 流式传输视频 449
 路径 276
 滤镜
 常见任务 293
 创建 294
 说明 295
 用于图像, 示例 313
 位图缓存和 296
 为显示对象删除 295
 用于显示对象和位图对象 300
 应用于 BitmapData 对象 295
 应用于显示对象 294
 在运行时更改 296
 逻辑运算符 60
M
 m 标志 (在正则表达式中) 180
Macintosh, FLV 文件 468
match() 方法 121
Matrix 类
 定义渐变 269
 对象, 定义 289
 平移 289
 倾斜 289
 示例 290
 缩放 289
 旋转 289
 已定义 288
Matrix3D 类 430
matrix3D 属性 425
 MAX_VALUE (Number 类) 47
method 属性 (URLRequest 类) 511
Microphone 类 213
milliseconds 属性 109
MIN_VALUE (Number 类) 47
minutes 属性 109
MIP map 413
mms.cfg 文件 590
month 属性 109
monthUTC 属性 109
MorphShape 类 229
Motion 类 351
MotionBase 类 349
MovieClip 对象, 创建 340
MovieClip 类 228
 帧速率 236
MouseEvent 类 210, 214
mx.util.Delegate 类 216
 麦克风
 安全性 606, 609
 访问 488
 检测活动 490
 冒号 (:) 运算符 43
 冒泡阶段 210
 枚举 85
 美元符号 (\$) 替换代码 122
 美元符号 (\$) 元字符 174
 密封类 46
 面向对象的编程
 常见任务 75
 概念 75
 名称冲突, 避免 33, 34
 命名空间
 AS3 100, 143
 打开 36
 导入 38
 定义 35, 77
flash_proxy 37
 访问控制说明符 36
 关于 35
 默认命名空间 35
namespace 关键字 35
use namespace 指令 36, 38, 100
 XML 201
 引用 36

应用 36

用户定义的属性 80

命名组（正则表达式中） 180

默认参数值 70

默认数据类型 31

默认行为

取消 212

已定义 210

目标节点或阶段 210

N

NaN 值 47

navigateToURL() 596

navigateToURL() 函数 597, 607

NetConnection 类 597

NetConnection.call() 596

NetConnection.connect() 596

NetConnection.connect() 方法 597, 599

NetStream 类 594, 597, 599

NetStream.play() 596

new 运算符 32

null 值 42, 47, 48, 140

Number 类

isNaN() 全局函数 42

精度 47

默认值 42

整数范围 47

转换 49

Number 数据类型 47

内存管理 140

内容，动态加载 257

内置类 31

匿名函数 66, 71

O

Object 类

关联数组 137

prototype 属性 97, 99

数据类型和 48

valueOf() 方法 100

on() 事件处理函数 209

onClipEvent() 函数 209

override 关键字 83, 84

P

package 语句 76

parentAllowsChild 属性 602

parse() 方法 82

PerspectiveProjection 类 427

Pixel Bender

关于 319

内核 319

在 ActionScript 中使用 319

着色器 319

术语 320

Pixel Bender 参数

默认值 324

Pixel Bender 元数据 322

Pixel Bender 着色器 319

背景处理 336

标识输入和参数 323

参数 322

访问元数据 322

非图像数据 336

嵌入 SWF 文件中 321

输入 322

用作绘制填充 327

用作滤镜 333

运行时加载 321

在 ActionScript 中使用 327

在独立模式下使用 336

指定输入值 323

字节代码 321

作为混和模式使用 330

Pixel Bender 着色器参数

顺序 326

play() 方法（NetStream 类） 597

player. 请参阅 Flash Player

PNG 图形 257

Point 对象

点间距离 285

关于 284

平移坐标空间 285

其它用途 285

polar() 方法 285

pop() 方法 133

preventDefault() 方法 210, 214

printArea 参数 565

PrintJob 语句，计时 565

PrintJob() 构造函数 563

priority 参数，addEventListener() 方法 218

private 属性 78

ProgressEvent.PROGRESS 510

projectionCenter 属性 427

protected 属性 79

__proto__ 31

prototype 属性 97, 99

Proxy 类 37

public 属性 78

push() 方法 132, 144

平滑位图 407

平移 423

平移矩阵 288

Q

奇偶规则 279

嵌入字体

使用 370

已定义 360

嵌入资源类 86

嵌套包 33

嵌套函数 69, 73

前缀运算符 58

倾斜矩阵 288, 289

倾斜显示对象 288

区分大小写 52

全局变量 40

全局对象 73

全局作用域 73

全屏

终止 448

全屏模式 237, 597

全屏视频 446

权限

LocalConnection 类 606

摄像头 464

R

Rectangle 对象

并集 287

打印 566

调整大小 286

交集 287

其它用途 288

已定义 286
 重新定位 286
RegExp 类
 方法 183
 关于 171
 属性 180
`replace()` 方法 112, 122
`_resolve` 31
`rest` 参数 71
`return` 语句 68, 82
`reverse()` 方法 134
`rotate()` 方法 289
`rotationX` 属性 426
`rotationY` 属性 426
`rotationZ` 属性 427
 RSS 数据
 读取播客频道 492
 加载, 示例 203
 RTMP 内容安全性 600
 日期和时间
 关于 108
 示例 108
 日期运算 110
 弱引用 140

S
s 标志 (在正则表达式中) 180
`sameDomain` 属性 602
`scale()` 方法 289
`search()` 方法 121
`seconds` 属性 109
Security 类 597
`Security.allowDomain()` 方法 595
 构造函数和 606
 关于跨脚本访问 600
 标签和 599
 加载上下文 259
 声音和 603
 `Security.currentDomain` 属性 606
 `Security.exactSettings` 属性 606
 `Security.loadPolicyFile()` 593, 595, 596
 `security.sandboxType` 属性 589
 `SecurityDomain` 类 259, 599
 `send()` 方法 (`LocalConnection` 类) 515, 597
 `sendToURL()` 596

`sendToURL()` 函数 597, 605
`setClipboard()` 方法 609
`setData()` 方法
 `Clipboard` 方法 559
`setDataHandler()` 方法 (`Clipboard` 类) 559
`setInterval()` 函数 113
`setter`。请参阅 `getter` 和 `setter`
`setTime()` 方法 110
`setTimeout()` 方法 113
Shader 类 321
 `data` 属性 322
ShaderData 类 322
ShaderFilter 类 333
ShaderInput 类 323
 `input` 属性 324
ShaderJob class
 `start()` 方法 336
ShaderJob 类 336
 `target` 属性 336
ShaderParameter 类 324
 `index` 属性 326
 `type` 属性 326
 `value` 属性 324
Shape 类 228
SharedObject 类 522, 597
`SharedObject.getLocal()` 方法 606, 608
`SharedObject.getRemote()` 方法 608
`shift()` 方法 133
SimpleButton 类 228
SimpleClock 示例 113
`slice()` 方法
 `String` 类 120
Socket 类 519, 597, 604
Sound 类 594, 597, 599
`Sound.load()` 596
SoundFacade 类 493
SoundLoaderContext 类 594
`SoundMixer.computeSpectrum()` 方法 600, 602, 603
`SoundMixer.stopAll()` 方法 603
`splice()` 方法 132, 133
`split()` 方法 121
Sprite 类 228
 `sprite`, 第一个加载的 225, 258

SpriteArranger 类示例 260
Stage
 安全性 601
Stage 类 210
StageDisplayState 类 597
static 属性 80
StaticText 类 229
`stopAll()` 方法 (`SoundMixer` 类) 603
`stopImmediatePropogation()` 方法 214
`stopPropogation()` 方法 214
String 类
 `charAt()` 方法 118
 `charCodeAt()` 方法 118
 `concat()` 方法 119
 `fromCharCode()` 方法 119
 `indexOf()` 方法 121
 `lastIndexOf()` 方法 121
 `match()` 方法 121
 `replace()` 方法 122
 `search()` 方法 121
 `slice()` 方法 120
 `split()` 方法 121
 `substr()` 和 `substring()` 方法 120
 `toLowerCase()` 和 `toUpperCase()` 方法 123
String 数据类型 47
StyleSheet 类 367
`substr()` 和 `substring()` 方法 120
SWF 文件
 导入加载的 606
 加载 257
 加载旧版本 344
 加载外部 343
 确定运行时环境 546
 实例之间的通信 517
 域之间的通信 518
`switch` 语句 62
`super` 语句 82, 93
`System.setClipboard()` 方法 609
 三角形, 绘制 433
 三维呈现 438
 三维旋转 436
 三元运算符 56
 沙箱 588
 上下文菜单, 自定义 503
 上载文件 529, 535, 605

- 上传 44
- 设备字体 360
- 摄像头
 - 安全性 606, 609
 - 播放条件 466
 - 捕获输入 462
 - 权限 464
 - 验证安装 463
 - 在屏幕上显示内容 462
- 深度管理, 改进 229
- 生成路径 34
- 声音
 - 安全性 599, 603
 - 范例应用程序 491
 - 向服务器发送和从中接收 491
- 事件
 - 请参阅事件侦听器
 - 安全性 602
 - 此关键字 216
 - 错误 160, 219
 - 调度 207, 219
 - enterFrame 事件 211
 - 父节点 211
 - 基本概念 11
 - init 事件 211
 - 默认行为 210
 - 目标节点 210
 - 事件对象 211
 - 事件流 207, 210, 212
 - 针对显示对象 240
 - 状态更改 162
 - 事件处理函数 209
 - 时间单位值 109
 - 事件对象 207
 - 时间格式 108
 - 时间间隔 110
 - 事件流 207, 210, 212
 - 事件目标 207, 210
 - 事件侦听器
 - 创建 214
 - 管理 217
 - 关于 207
 - 类的外部 215
 - 作为类方法 215
 - 删除 218
- 要避免的技术 216
- 在 ActionScript 3.0 中的变化 210
- 时间轴, Flash 20
 - 示例
 - 错误处理 219
 - 多页打印 567
 - GeometricShapes 101
 - 构建 Telnet 客户端 536
 - 过滤图像 313
 - 加载 RSS 数据 203
 - 检测系统功能 553
 - 将外部 API 用于网页容器 576
 - Matrix 类 290
 - RunTimeAssetsExplorer 344
 - SimpleClock 113
 - SpriteArranger 类 260
 - 声音应用程序 491
 - 视频自动唱片点唱机 468
 - 数组 145
 - Wiki 分析器 184
 - WordSearch 504
 - 文本格式设置 372
 - 正则表达式 184
 - 重新排列显示对象层 263
 - 字符串 123
 - 实例, 创建 15
 - 实例变量 81
 - 实例方法 82
 - 实例属性
 - 继承 90
 - 声明 78
 - 矢量, 3D 423
 - 矢量打印 565
 - 视频
 - 安全性 599, 603
 - 播放 444
 - 常见任务 440
 - Flash Player 和 AIR 兼容性 442
 - FLV 格式 443
 - 发送到服务器 467
 - fullScreenSourceRect 属性 447
 - 关于 440
 - H.264 支持 442
 - 加载 444
 - 理解格式 442
- 流 449
- 流末尾 445
- 在 Macintosh 上 468
- mipmap 处理 413
- 品质 465
- 使用全屏 446
- 使用提示点和元数据 454
- 硬件加速用于全屏 449
- 元数据 455, 458
- 终止全屏 448
- 视频自动唱片点唱机示例 468
- 时区 109, 110
- 实时消息传递协议内容安全性 600
- 使影片剪辑后退 339
- 使影片剪辑快进 339
- 时钟示例 113
- 鼠标安全性 609
- 鼠标光标, 自定义 502
- 数据
 - 安全性 602, 605
 - 发送到服务器 514
 - 加载外部 510
 - 数据格式, 剪贴板 558
 - 数据结构 128
 - 数据类型
 - Boolean 46
 - 关于 9
 - int 47
 - 简单和复杂 9
 - 默认 (无类型) 31
 - Number 47
 - String 47
 - uint 47
 - void 48
 - 已定义 42
 - 自定义 85
 - 数量表示符 (正则表达式中) 177
- 输入文本字段 360
- 竖线 (|) 字符 178
- 属性
 - ActionScript 与其它语言 31
 - 基本概念 10
 - 静态和实例 78, 94
 - 添加到函数 73
 - XML 192

- 已定义, 为 ActionScript 3.0 78
 正则表达式 180
 属性标识符 (@) 运算符, XML 191, 199
 属性访问运算符 138
 数组
 插入元素 132
 查询 136
 长度 134
 常见任务 128
 超类构造函数 143
 创建 121, 131
 对象键 138
 多维 140
 关联 137
 关于 128
 键和值对 137
 检索值 133
 克隆 142
 排序 134
 浅副本 142
 嵌套数组和 join() 方法 137
 删除元素 133
 删除运算符 133
 深副本 142
 示例 145
 使用关联数组和索引数组 141
 数组文本 53, 131
 索引 129
 循环访问 139
 术语 129
 最大大小 129
 数组的超类构造函数 143
 数组访问运算符 130
 数组中的对象键 138
 私有类 32
 搜索, 正则表达式中 182
 搜索字符串 121
 速度, 提高呈现的 249
 缩放
 打印 566
 矩阵 288
 控制扭曲 247
 舞台 236
 显示对象 288
 缩放值, 透视 434
 索引数组 129
- T**
- T 缩放值 434
 T 值 424
 tailjoint 属性 356
 target 属性 213
 Telnet 客户端示例 536
 test() 方法 183
 TextEvent 类 210
 TextField 类 210, 228
 复制和粘贴 557
 TextFormat 类 367
 TextLineMetrics 类 378
 TextSnapshot 类 372
 this 关键字 82, 83, 84, 216
 throw 语句 156
 Timer 类
 关于 111
 监视播放 496
 timer 事件 111
 toLowerCase() 方法 123
 toString() 方法
 Event 类 214
 关于 119
 toUpperCase() 方法 123
 traits 对象 99
 Transform 类
 transform 属性 289
 translate() 方法 289
 TriangleCulling 439
 try..catch..finally 语句 155
 type 参数 131
 type 属性 (Event 类) 212
 套接字策略文件 592, 604
 套接字服务器 520
 套接字连接 519, 604
 缇 566
 剔除 424, 438
 替换代码 122
 替换字符串中的文本 121
 提升 41
- 提示点
 视频中 449
 使用 454
 填充规则 278
 条件 (?) 运算符 60
 条件语句 61
 停止影片剪辑 339
 同步错误 152
 通配符 (*) 运算符, XML 199
 通信
 不同域中的 SWF 文件之间 518
 在 Flash Player 实例之间 515
 在 SWF 文件之间 517
 统一资源标识符 (URI) 35
 通用对象 53, 137
 通用时间 (UTC) 108
 通用协调时间 (UTC) 108
 透视 423, 434
 投影 423
 图像
 安全性 602
 过滤示例 313
 加载 257
 文本字段中 362
 在 Bitmap 类中定义 228
 图形, 加载 257
 图形数据类 280
 拖放
 捕获交互 501
 创建交互 242
- W**
- valueOf() 方法 (Object 类) 100
 var 关键字 39, 80
 Vector 的优点 130
 Vector 基本类型 130
 Vector 类
 concat() 方法 136
 创建固定长度 Vector 132
 创建实例 131
 构造函数 131
 关于 130
 join() 方法 136
 reverse() 方法 134
 slice() 方法 136

sort() 方法 135
toString() 方法 136
Vector 全局函数 132
Vector 限制 130
Vector3D 类 430
vertices 433
while 循环 65
Video 类 444
UIEventDispatcher 类 209
 Wiki 分析器示例 184
willTrigger() 方法 219
uint 类, 转换 49
uint 数据类型 47, 48
undefined 32, 48
Unicode 字符 116
union() 方法 287
unshift() 方法 132
void 48
WordSearch 示例 504
upload() 方法 597, 605
URI 35
URL
 复制和粘贴支持 557
URL 编码 511
URL 策略文件 592
URLLoader 构造函数 510
URLLoader 类
 安全性和 604
 关于 510
 加载 XML 数据 195, 203
 受限制时 597
URLLoader.dataFormat 属性 514
URLLoader.load() 596
URLLoader.load() 方法 510, 511
URLLoaderDataFormat.VARIABLES 514
URLRequest 实例 510, 511
URLRequest.contentType 属性 510
URLRequest.data 属性 511
URLRequest.method 属性 511
URLRequestMethod.GET 511
URLRequestMethod.POST 511
URLStream 类 597, 604
URLStream.load() 596
URLVariables 类 510
URLVariables.decode() 方法 511

use namespace 指令 36, 38, 100
useCapture 参数, **addEventListener()** 方法 218
useWeakReference 参数 140
UTC (通用协调时间) 108
UV 映射 424, 434
外部 API
 常见任务 571
 概念和术语 571
 关于 571
 示例 576
 XML 格式 575
 优点 573
 外部 SWF 文件, 加载 343
 外部代码, 从 ActionScript 中调用 574
 外部容器, 获取相关信息 574
 外部数据, 加载 510
 外部文档, 加载数据 511
 外观类 493
网络
 概念和术语 508
 关于 508
 限制 595
 网络字节顺序 519
 尾数 47
位图
 安全性 602
 复制和粘贴支持 557
 关于 405
 平滑处理 407
 透明与不透明 406
 文件格式 405
 优化 413
 在 Bitmap 类中定义 228
位图打印 565
位图缓存
 何时避免 249
 何时使用 249
 滤镜和 296
 优点和缺点 249
位图数据, 复制 410
位图转换 434
位置
 显示对象的 241
 字符在字符串中的 121

文本
 保存到剪贴板中 546
 捕获输入 365
操作 364
常见任务 359
 粗细 371
 复制和粘贴支持 557
概念和术语 360
 关于 361
 滚动 363, 364
静态 229, 372
可用类型 361
清晰度 371
设置范围格式 369
设置格式 367, 372
替换 121
显示 361
限制输入 366
消除锯齿 371
选择 364
指定格式 367
文本编辑器 22
文本行量度 360, 378
文本引擎
 处理事件 384
 创建和显示文本 380
 创建行 380
 对齐东亚文字 393
 对齐文本 392
ElementFormat 380
ElementFormat 对象 387
ElementFormat 锁定 / 克隆 389
FontDescription 390
FontDescription 锁定 / 克隆 391
fontLookup 属性 391
fontName 属性 390
fontPosture 属性 390
GraphicElement 380, 381
GroupElement 380, 382
换行符 395
换行文本 395
基线对齐 387
基线偏移 388
间距 395
镜像事件 385

- 控制文本 392
- 嵌入字体和设备字体 391
- replaceText 方法 383
- 设置文本格式 386
- 使用字体 390
- Tab 停靠位 395
- TextBlock 380
- TextElement 380
- TextLine 380
- 替换文本 383
- 添加图形 381
- 文本大小写 388
- 新闻布局示例 396
- 旋转文本 388
- 亚洲文字支持 393
- 亚洲语言支持 388
- 字距调整 394
- 字母间距 393
- 字体呈现 391
- 字体提示 391
- 字体透明度 (alpha) 387
- 字体颜色 387
- 文本字段
 - 动态 360
 - 滚动文本 363
 - HTML 367
 - img 标签和安全性 599
 - 禁用 IME 551
 - 静态 360
 - 输入 360
 - 图像 362
 - 修改 361
- 文本字段中的 img 标签, 安全性 599
- 文档
 - ActionScript 1
 - ActionScript 3.0 编程内容 1
 - Adobe 开发人员中心和 Adobe 设计中心 2
 - Flash 2
- 文档对象模型 (DOM) 第 3 级事件规范 207, 209
- 问号 (?) 元字符 174
- 文件
 - 保存 527
 - 复制和粘贴支持 557
 - 加载 526
- 上传 535, 605
- 下载 605
- 文件大小, 形状的较小 229
- 纹理式网格 423
- 无类型变量 32, 42
- 无穷大 47
- 舞台
 - 关于 210, 225
 - 属性, 设置 236
 - 缩放 236
 - 作为显示对象容器 226
 - 舞台所有者 601
 - 无序数组 137
- X**
 - x 标志 (在正则表达式中) 180
 - XML
 - ActionScript 189
 - 遍历结构 197
 - 常见任务 190
 - 处理指令 192
 - 初始化变量 195
 - 大括号运算符 ({ 和 }) 196
 - E4X (ECMAScript for XML) 33, 188, 191
 - for each..in 循环 64
 - for 循环 191, 200
 - 方法 192
 - 访问属性 199
 - 父节点 198
 - 概念和术语 190
 - 过滤 199
 - 基本概念 188
 - 加载数据 195, 203
 - 空白 192
 - 类型转换 202
 - 命名空间 201
 - 属性 192
 - 套接字服务器 520
 - 外部 API 的格式 575
 - 文档 189
 - 注释 192
 - 转换 196
 - 子节点 198
 - XML 类 33
 - XML 中的大括号运算符 ({ 和 }) 196
 - XML 中的节点, 访问 198
 - XMLDocument 类 33, 191
 - XMLList 对象
 - 关于 194
 - 连接 197
 - XmlNode 类 191
 - XMLParser 类 191
 - XMLSocket 类 196, 203, 520, 597, 604
 - XMLSocket.connect() 方法 597
 - XMLTag 类 191
 - 系统, 确定用户的 545
 - 下载文件 534, 605
 - 显示编程, 关于 224
 - 显示对象
 - 安全性 602
 - 常见任务 226
 - 创建 231
 - 大小 246
 - 淡化 253
 - 单击并拖动示例 263
 - 调整颜色 251
 - 定位 241
 - 动画处理 256
 - 关于 225
 - 过滤 293, 294, 300
 - 核心类继承 228
 - 缓存 248
 - 绘图 API 265
 - 矩阵转换 289
 - 列表外 230
 - 平移 288
 - 倾斜 288
 - 删除滤镜 295
 - 设置颜色 251
 - 深度管理 229
 - 事件 240
 - 示例 259, 274
 - 缩放 246, 247, 288
 - 添加到显示列表 231
 - 位图 405
 - 选择子类 240
 - 旋转 253, 288
 - 影片剪辑 337
 - 用户输入和 498
 - 遮罩 254

- 重新排列示例 263
 术语 226
 子类化 230
 组合 232
 组合复杂的对象 230
 显示对象容器 225, 232
 显式类型转换 48
 显示列表
 安全性 602
 遍历 235
 关于 224
 事件流 210
 优点 229
 显示列表对象 209
 显示内容, 动态加载 257
 显示器, 全屏模式 237
 显示体系结构 224, 265
 像素, 处理单个 408
 像素级别冲突检测 409
 像素贴紧 407
 消除文本锯齿 371
 小括号
 空 66
 元字符 174
 运算符 53
 消失点 423
 小于或等于运算符 119
 小于运算符 57, 119
 斜杠
 反斜杠 (\) 174
 反斜杠 (\\\) 117
 正斜杠 (/) 173, 174
 斜杠语法 52
 星号 (*) 类型注释 42, 44, 48
 星号 (*) 元字符 174
 星号 (*)。请参阅星号
 星号 (通配符) 运算符, XML 199
 性能, 提高显示对象的 248
 形状, 绘制 277
 序列化对象
 复制和粘贴支持 557
 旋转 423
 旋转矩阵 288
 旋转显示对象 253, 288
- 循环
 do..while 65
 for 63
 for (XML) 191, 200
 for each..in 64, 139, 200
 for..in 63, 139, 200
 while 65
 循环访问数组 139
- Y**
 延迟呈现 (复制和粘贴) 559
 严格模式
 点语法和 67
 返回值 68
 关于 43
 显式转换 49
 运行时错误 44
 转换 49
- 颜色
 背景 250
 更改特定的 252
 设置显示对象的 251
 在显示对象中调整 251
 组合不同图像中的 250
- 扬声器
 传送到本地扬声器 489
 扬声器和麦克风 489
 样式表. 请参阅 CSS
 页面属性 565
 异步操作 219
 异步错误 152
 异常 152
 一元运算符 56, 58
 以字符分隔的字符串, 将数组合并到 148
 引号 117
 音频安全性 603
 音频回放, 监视 496
 音频回放进度 496
 隐式类型转换 48
 引用, 传递方式 69
 硬件加速, 全屏 449
 硬件缩放 239
 影片剪辑
 播放和停止 339
 常见任务 337
- 概念和术语 337
 关于 337
 后退 339
 快进 339
 帧速率 236
 映射 137, 138
 应用程序, 开发决策 20
 用户的系统, 在运行时确定 545
 用户交互, 管理焦点 503
 用户输入
 常见任务 498
 概念和术语 498
 关于 498
 用户选择的文本, 捕获 365
 右键单击菜单 (上下文菜单) 503
 右结合的运算符 56
 右小括号 174
 有效位数 47
 右中括号 174
 右中括号 () 174
 域, 之间的通信 518
 语法 52
 元策略 593
 源路径 34
 元数据, 视频 455, 458
 使用 454
 原型对象 67, 97, 99
 原型链 31, 97
 元序列, 正则表达式中 174, 175
 元字符, 正则表达式中 174
 运算符
 按位逻辑 60
 按位移位 59
 乘法 59
 等于 60, 119
 赋值 61
 关系 59
 关于 56
 后缀 58
 基本概念 16
 加法 59
 逻辑 60
 前缀 58
 条件 60
 一元 56, 58

- 优先级 56
- 主要 57
- 运行时, 确定用户的系统 545
- Z**
- z 属性** 425
- 在屏幕上显示摄像头内容 462
- 着色器混和模式 330
- 着色器滤镜 333
- 遮蔽 95
- 遮罩显示对象 254
- 帧, 跳到 339
- 侦听器, 请参阅事件侦听器
- 正无穷大 47
- 正斜杠 173, 174
- 正则表达式
 - 标志 180
 - 捕获子字符串匹配 179
 - 创建 173
 - 关于 171
 - 逻辑“或”使用竖线 (|) 元字符 178
 - 逻辑“或”字符和字符组 179
 - 命名组 180
 - String 方法中的参数 184
 - 示例 184
 - 使用方法 183
 - 数量表示符 177
 - 属性 180
 - 搜索 182
 - 元序列 174, 175
 - 元字符 174
 - 正斜杠分隔符 173
 - 字符 174
 - 字符类 176
 - 组 178
 - 正则表达式的 dotall 属性 180
 - 正则表达式的 extended 属性 180
 - 正则表达式的 ignoreCase 属性 180
 - 正则表达式的 multiline 属性 180
 - 正则表达式的全局属性 180
 - 正则表达式中的 dotall 标志 182
 - 正则表达式中的 extended 标志 182
 - 正则表达式中的 global 标志 181
 - 正则表达式中的 ignore 标志 181
 - 正则表达式中的 multiline 标志 181
 - 正则表达式中的标志 180
 - 正则表达式中的非捕获组 180
 - 正则表达式中的符号 174
 - 正则表达式中的逻辑“或” 178
 - 值
 - 为变量赋值 39
 - 传递参数方式 69
 - 制表符 117
 - 指定类型的数组 130
 - 指针 (光标), 自定义 502
 - 中括号 ([和]) 字符 174
 - 重载运算符 56
 - 主策略文件 593
 - 注释
 - 关于 17, 54
 - 在 XML 中 192
 - 主要运算符 57
 - 转换 48, 49, 50, 423
 - 转换矩阵。请参阅 Matrix 类
 - 状态更改事件 162
 - 字, 保留 54
 - 自定义 LocalConnection 客户端 516
 - 自定义错误类 159
 - 自定义类 23
 - 自定义数据类型, 枚举 85
 - 字符
 - 在正则表达式中 174
 - 在字符串中 118, 121
 - 字符串
 - 比较 119
 - 查找子字符串 120
 - 长度 118
 - 常见任务 116
 - 检查正则表达式中的匹配内容 183
 - 将 XML 对象转换为 202
 - 将数组合并为以字符分隔的字符串 148
 - 连接 119
 - 模式, 查找 120, 121
 - 匹配子字符串 179
 - 声明 117
 - 示例 123
 - 索引位置 118
 - 替换文本 121
 - 术语 116
 - 转换 XML 属性的数据类型 203
 - 转换大小写 123
 - 字符位置 121
 - 子字符串 120, 121
 - 字符串键 137
 - 字符串中的大小写替换 123
 - 字符串中的单引号 117
 - 字符串中的双引号 117
 - 字符串中的索引位置 118
 - 字符代码 500
 - 字符范围, 指定 177
 - 字符类 (在正则表达式中) 176
 - 字符类中的转义序列 176
 - 字节顺序 519
 - 子类 89
 - 字面值
 - 对象 137
 - 关于 53
 - 数组文本 53, 131
 - 字体
 - 嵌入 360, 370
 - 设备 360
 - 子字符串
 - 查找和替换 120, 121
 - 关于 120
 - 基于分隔符创建 121
 - 在正则表达式中匹配 179
 - 纵向打印 567
 - 组, 正则表达式中 178
 - 组合显示对象 232
 - 坐标空间
 - 平移 285
 - 已定义 283
 - 坐标系, 3D 424
 - 左结合的运算符 56
 - 左小括号 174
 - 作用域链 73, 95
 - 左中括号 174
 - 左中括号 ([) 174